Rules for integrands of the form $(a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n$

1. $\int (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx$

   1. $\int (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx$ when $a^2 - b^2 - c^2 == 0$

      1. $\int (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx$ when $a^2 - b^2 - c^2 == 0 \land n > 0$

         1: $\int \sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]} \, dx$ when $a^2 - b^2 - c^2 == 0$

- Reference: G&R 2.558.1 inverted with $n == \frac{1}{2}$ and $a^2 - b^2 - c^2 == 0$
- Rule: If $a^2 - b^2 - c^2 == 0$, then

$$\int \sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]} \, dx \longrightarrow -\frac{2 (c \operatorname{Cos}[d + e x] - b \operatorname{Sin}[d + e x])}{e \sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]}}$$

- Program code:

```
Int[Sqrt[a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]],x_Symbol] :=
  -2*(c*Cos[d+e*x]-b*Sin[d+e*x])/(e*Sqrt[a+b*Cos[d+e*x]+c*Sin[d+e*x]]) /;
FreeQ[{a,b,c,d,e},x] && EqQ[a^2-b^2-c^2,0]
```

         2: $\int (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx$ when $a^2 - b^2 - c^2 == 0 \land n > 1$

- Reference: G&R 2.558.1 inverted with $a^2 - b^2 - c^2 == 0$
- Rule: If $a^2 - b^2 - c^2 == 0 \land n > 0$, then

$$\int (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx \longrightarrow$$

$$-\frac{(c \operatorname{Cos}[d + e x] - b \operatorname{Sin}[d + e x]) (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^{n-1}}{e n} + \frac{a (2 n - 1)}{n} \int (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^{n-1} \, dx$$

- Program code:

```
Int[(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_,x_Symbol] :=
  -(c*Cos[d+e*x]-b*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-1)/(e*n) +
  a*(2*n-1)/n*Int[(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[a^2-b^2-c^2,0] && GtQ[n,0]
```

2. $\int (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx$ when $a^2 - b^2 - c^2 == 0 \land n < 0$

1: $\int \frac{1}{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]} \, dx$ when $a^2 - b^2 - c^2 == 0$

Reference: G&R 2.558.4d

Rule: If $a^2 - b^2 - c^2 == 0$, then

$$\int \frac{1}{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]} \, dx \to -\frac{c - a \operatorname{Sin}[d + e x]}{c e (c \operatorname{Cos}[d + e x] - b \operatorname{Sin}[d + e x])}$$

Program code:

```
Int[1/(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  -(c-a*Sin[d+e*x])/(c*e*(c*Cos[d+e*x]-b*Sin[d+e*x])) /;
FreeQ[{a,b,c,d,e},x] && EqQ[a^2-b^2-c^2,0]
```

2: $\int \frac{1}{\sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]}} \, dx$ when $a^2 - b^2 - c^2 == 0$

Derivation: Algebraic simplification

- **Basis: If $a^2 - b^2 - c^2 == 0$, then $a + b \operatorname{Cos}[z] + c \operatorname{Sin}[z] == a + \sqrt{b^2 + c^2} \operatorname{Cos}[z - \operatorname{ArcTan}[b, c]]$**

Rule: If $a^2 - b^2 - c^2 == 0$, then

$$\int \frac{1}{\sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]}} \, dx \to \int \frac{1}{\sqrt{a + \sqrt{b^2 + c^2} \operatorname{Cos}[d + e x - \operatorname{ArcTan}[b, c]]}} \, dx$$

Program code:

```
Int[1/Sqrt[a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]],x_Symbol] :=
  Int[1/Sqrt[a+Sqrt[b^2+c^2]*Cos[d+e*x-ArcTan[b,c]]],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[a^2-b^2-c^2,0]
```

3: $\int (a + b \cos[d + e x] + c \sin[d + e x])^n \, dx$ when $a^2 - b^2 - c^2 == 0 \land n < -1$

Reference: G&R 2.558.1 inverted with $a^2 - b^2 - c^2 == 0$ inverted

Rule: If $a^2 - b^2 - c^2 == 0 \land n < -1$, then

$$\int (a + b \cos[d + e x] + c \sin[d + e x])^n \, dx \longrightarrow$$
$$\frac{(c \cos[d + e x] - b \sin[d + e x]) \, (a + b \cos[d + e x] + c \sin[d + e x])^n}{a e \, (2 n + 1)} + \frac{n + 1}{a \, (2 n + 1)} \int (a + b \cos[d + e x] + c \sin[d + e x])^{n+1} \, dx$$

Program code:

```
Int[(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_,x_Symbol] :=
  (c*Cos[d+e*x]-b*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n/(a*e*(2*n+1)) +
  (n+1)/(a*(2*n+1))*Int[(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[a^2-b^2-c^2,0] && LtQ[n,-1]
```

2. $\int (a + b \cos[d + e x] + c \sin[d + e x])^n \, dx$ when $a^2 - b^2 - c^2 \neq 0$

1. $\int (a + b \cos[d + e x] + c \sin[d + e x])^n \, dx$ when $a^2 - b^2 - c^2 \neq 0 \land n > 0$

1. $\int \sqrt{a + b \cos[d + e x] + c \sin[d + e x]} \, dx$ when $a^2 - b^2 - c^2 \neq 0$

1: $\int \sqrt{a + b \cos[d + e x] + c \sin[d + e x]} \, dx$ when $b^2 + c^2 == 0$

Reference: Integration by substitution

Basis: If $b^2 + c^2 == 0$, then $f[b \cos[d + e x] + c \sin[d + e x]] == \frac{b \, f[b \cos[d+e x]+c \sin[d+e x]]}{c e \, (b \cos[d+e x]+c \sin[d+e x])} \partial_x (b \cos[d + e x] + c \sin[d + e x])$

Rule: If $b^2 + c^2 == 0$, then

$$\int \sqrt{a + b \cos[d + e x] + c \sin[d + e x]} \, dx \longrightarrow \frac{b}{c e} \operatorname{Subst}\left[\int \frac{\sqrt{a + x}}{x} \, dx, \, x, \, b \cos[d + e x] + c \sin[d + e x]\right]$$

Program code:

```
Int[Sqrt[a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]],x_Symbol] :=
  b/(c*e)*Subst[Int[Sqrt[a+x]/x,x],x,b*Cos[d+e*x]+c*Sin[d+e*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[b^2+c^2,0]
```

2. $\int \sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]} \, dx$ when $a^2 - b^2 - c^2 \neq 0 \land b^2 + c^2 \neq 0$

1: $\int \sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]} \, dx$ when $b^2 + c^2 \neq 0 \bigwedge a + \sqrt{b^2 + c^2} > 0$

Derivation: Algebraic simplification

- **Basis: If** $b^2 + c^2 \neq 0$**, then** $a + b \operatorname{Cos}[z] + c \operatorname{Sin}[z] == a + \sqrt{b^2 + c^2} \operatorname{Cos}[z - \operatorname{ArcTan}[b, c]]$

- **Rule: If** $b^2 + c^2 \neq 0 \bigwedge a + \sqrt{b^2 + c^2} > 0$**, then**

$$\int \sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]} \, dx \longrightarrow \int \sqrt{a + \sqrt{b^2 + c^2} \operatorname{Cos}[d + e x - \operatorname{ArcTan}[b, c]]} \, dx$$

Program code:

```
Int[Sqrt[a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]],x_Symbol] :=
  Int[Sqrt[a+Sqrt[b^2+c^2]*Cos[d+e*x-ArcTan[b,c]]],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2+c^2,0] && GtQ[a+Sqrt[b^2+c^2],0]
```

$$2:\ \int\sqrt{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}\,dx \text{ when } a^2-b^2-c^2\neq 0\ \bigwedge\ b^2+c^2\neq 0\ \bigwedge\ \neg\left(a+\sqrt{b^2+c^2}>0\right)$$

Derivation: Piecewise constant extraction and algebraic simplification

- **Basis:** $\partial_x \frac{\sqrt{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}}{\sqrt{\frac{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}{a+\sqrt{b^2+c^2}}}} == 0$
- **Basis: If** $b^2+c^2\neq 0$**, then** $a+b\,\mathrm{Cos}[z]+c\,\mathrm{Sin}[z] == a+\sqrt{b^2+c^2}\,\mathrm{Cos}[z-\mathrm{ArcTan}[b,\,c]]$
- **Rule: If** $a^2-b^2-c^2\neq 0\ \bigwedge\ b^2+c^2\neq 0\ \bigwedge\ \neg\left(a+\sqrt{b^2+c^2}>0\right)$**, then**

$$\int\sqrt{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}\,dx \longrightarrow$$

$$\frac{\sqrt{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}}{\sqrt{\frac{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}{a+\sqrt{b^2+c^2}}}}\int\sqrt{\frac{a}{a+\sqrt{b^2+c^2}}+\frac{\sqrt{b^2+c^2}}{a+\sqrt{b^2+c^2}}\,\mathrm{Cos}[d+e\,x-\mathrm{ArcTan}[b,\,c]]}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]],x_Symbol] :=
  Sqrt[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/Sqrt[(a+b*Cos[d+e*x]+c*Sin[d+e*x])/(a+Sqrt[b^2+c^2])]*
    Int[Sqrt[a/(a+Sqrt[b^2+c^2])+Sqrt[b^2+c^2]/(a+Sqrt[b^2+c^2])*Cos[d+e*x-ArcTan[b,c]]],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[a^2-b^2-c^2,0] && NeQ[b^2+c^2,0] && Not[GtQ[a+Sqrt[b^2+c^2],0]]
```

2: $\int (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx$ when $a^2 - b^2 - c^2 \neq 0 \wedge n > 1$

Reference: G&R 2.558.1 inverted

Rule: If $a^2 - b^2 - c^2 \neq 0 \wedge n > 1$, then

$$\int (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx \longrightarrow$$

$$-\frac{(c \operatorname{Cos}[d + e x] - b \operatorname{Sin}[d + e x]) (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^{n-1}}{e n} +$$

$$\frac{1}{n} \int \left(n a^2 + (n-1) \left(b^2 + c^2\right) + a b (2 n - 1) \operatorname{Cos}[d + e x] + a c (2 n - 1) \operatorname{Sin}[d + e x]\right) (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^{n-2} \, dx$$

Program code:

```
Int[(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_,x_Symbol] :=
  -(c*Cos[d+e*x]-b*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-1)/(e*n) +
  1/n*Int[Simp[n*a^2+(n-1)*(b^2+c^2)+a*b*(2*n-1)*Cos[d+e*x]+a*c*(2*n-1)*Sin[d+e*x],x]*
    (a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-2),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[a^2-b^2-c^2,0] && GtQ[n,1]
```

2. $\int(a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x])^n\,dx$ when $a^2-b^2-c^2\neq0\,\wedge\,n<0$

   1. $\int\frac{1}{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}\,dx$ when $a^2-b^2-c^2\neq0$

      x: $\int\frac{1}{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}\,dx$ when $a^2-b^2-c^2>0$

- **Note: Although this rule produces a more complicated antiderivative than the following rule, it is continuous provided $a^2-b^2-c^2>0$.**

- **Rule: If $a^2-b^2-c^2>0$, then**

$$\int\frac{1}{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}\,dx\rightarrow\frac{x}{\sqrt{a^2-b^2-c^2}}+\frac{2}{e\sqrt{a^2-b^2-c^2}}\,\text{ArcTan}\Big[\frac{c\,\text{Cos}[d+e\,x]-b\,\text{Sin}[d+e\,x]}{a+\sqrt{a^2-b^2-c^2}+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}\Big]$$

- **Program code:**

```
(* Int[1/(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  x/Sqrt[a^2-b^2-c^2] +
  2/(e*Sqrt[a^2-b^2-c^2])*ArcTan[(c*Cos[d+e*x]-b*Sin[d+e*x])/(a+Sqrt[a^2-b^2-c^2]+b*Cos[d+e*x]+c*Sin[d+e*x])] /;
FreeQ[{a,b,c,d,e},x] && GtQ[a^2-b^2-c^2,0] *)
```

x: $\int \frac{1}{a + b \cos[d + e x] + c \sin[d + e x]} \, dx$ when $a^2 - b^2 - c^2 < 0$

- **Note: Although this rule produces a more complicated antiderivative than the following rule, it is continuous provided $a^2 - b^2 - c^2 < 0$.**

Rule: If $a^2 - b^2 - c^2 < 0$, then

$$\int \frac{1}{a + b \cos[d + e x] + c \sin[d + e x]} \, dx \longrightarrow$$

$$\text{Log}\left[b^2 + c^2 + \left(a b - c \sqrt{-a^2 + b^2 + c^2}\right) \cos[d + e x] + \left(a c + b \sqrt{-a^2 + b^2 + c^2}\right) \sin[d + e x]\right] \Big/ \left(2 e \sqrt{-a^2 + b^2 + c^2}\right) -$$
$$\frac{1}{2 e \sqrt{-a^2 + b^2 + c^2}} \text{Log}\left[b^2 + c^2 + \left(a b + c \sqrt{-a^2 + b^2 + c^2}\right) \cos[d + e x] + \left(a c - b \sqrt{-a^2 + b^2 + c^2}\right) \sin[d + e x]\right]$$

- **Program code:**

```
(* Int[1/(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  Log[RemoveContent[b^2+c^2+(a*b-c*Rt[-a^2+b^2+c^2,2])*Cos[d+e*x]+(a*c+b*Sqrt[-a^2+b^2+c^2])*Sin[d+e*x],x]]/
    (2*e*Rt[-a^2+b^2+c^2,2]) -
  Log[RemoveContent[b^2+c^2+(a*b+c*Rt[-a^2+b^2+c^2,2])*Cos[d+e*x]+(a*c-b*Sqrt[-a^2+b^2+c^2])*Sin[d+e*x],x]]/
    (2*e*Rt[-a^2+b^2+c^2,2]) /;
FreeQ[{a,b,c,d,e},x] && LtQ[a^2-b^2-c^2,0] *)
```

1: $\int \frac{1}{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}\,dx$ when $a+b==0$

Derivation: Integration by substitution

Basis: $\frac{1}{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]} == -\frac{2}{e}\,\text{Subst}\left[\frac{1}{a-b+2\,c\,x+(a+b)\,x^2},\,x,\,\text{Cot}\left[\frac{1}{2}\,(d+e\,x)\right]\right]\,\partial_x\text{Cot}\left[\frac{1}{2}\,(d+e\,x)\right]$

Basis: If $a+b==0$**, then** $\frac{1}{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]} == -\frac{1}{e}\,\text{Subst}\left[\frac{1}{a+c\,x},\,x,\,\text{Cot}\left[\frac{1}{2}\,(d+e\,x)\right]\right]\,\partial_x\text{Cot}\left[\frac{1}{2}\,(d+e\,x)\right]$

Rule: If $a+b==0$**, then**

$$\int \frac{1}{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}\,dx \rightarrow -\frac{1}{e}\,\text{Subst}\left[\int \frac{1}{a+c\,x}\,dx,\,x,\,\text{Cot}\left[\frac{1}{2}\,(d+e\,x)\right]\right]$$

Program code:

```
Int[1/(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  Module[{f=FreeFactors[Cot[(d+e*x)/2],x]},
  -f/e*Subst[Int[1/(a+c*f*x),x],x,Cot[(d+e*x)/2]/f]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[a+b,0]
```

2: $\int \frac{1}{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}\,dx$ when $a+c==0$

Derivation: Integration by substitution

Basis: $\frac{1}{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]} == \frac{2}{e}\,\text{Subst}\left[\frac{1}{a-c+2\,b\,x+(a+c)\,x^2},\,x,\,\text{Tan}\left[\frac{1}{2}\,(d+e\,x)+\frac{\pi}{4}\right]\right]\,\partial_x\text{Tan}\left[\frac{1}{2}\,(d+e\,x)+\frac{\pi}{4}\right]$

Basis: If $a+c==0$**, then** $\frac{1}{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]} == \frac{1}{e}\,\text{Subst}\left[\frac{1}{a+b\,x},\,x,\,\text{Tan}\left[\frac{1}{2}\,(d+e\,x)+\frac{\pi}{4}\right]\right]\,\partial_x\text{Tan}\left[\frac{1}{2}\,(d+e\,x)+\frac{\pi}{4}\right]$

Rule: If $a+c==0$**, then**

$$\int \frac{1}{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}\,dx \rightarrow \frac{1}{e}\,\text{Subst}\left[\int \frac{1}{a+b\,x}\,dx,\,x,\,\text{Tan}\left[\frac{1}{2}\,(d+e\,x)+\frac{\pi}{4}\right]\right]$$

Program code:

```
Int[1/(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  Module[{f=FreeFactors[Tan[(d+e*x)/2+Pi/4],x]},
  f/e*Subst[Int[1/(a+b*f*x),x],x,Tan[(d+e*x)/2+Pi/4]/f]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[a+c,0]
```

3: $\int \frac{1}{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]} \, dx$ when $a - c == 0$

Derivation: Integration by substitution

- **Basis:** $\frac{1}{a+b \operatorname{Cos}[d+e x]+c \operatorname{Sin}[d+e x]} == -\frac{2}{e} \operatorname{Subst}\left[\frac{1}{a+c+2 b x+(a-c) x^2}, x, \operatorname{Cot}\left[\frac{1}{2} (d + e x) + \frac{\pi}{4}\right]\right] \partial_x \operatorname{Cot}\left[\frac{1}{2} (d + e x) + \frac{\pi}{4}\right]$

- **Basis: If** $a - c == 0$, **then** $\frac{1}{a+b \operatorname{Cos}[d+e x]+c \operatorname{Sin}[d+e x]} == -\frac{1}{e} \operatorname{Subst}\left[\frac{1}{a+b x}, x, \operatorname{Cot}\left[\frac{1}{2} (d + e x) + \frac{\pi}{4}\right]\right] \partial_x \operatorname{Cot}\left[\frac{1}{2} (d + e x) + \frac{\pi}{4}\right]$

Rule: If $a - c == 0$, **then**

$$\int \frac{1}{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]} \, dx \longrightarrow -\frac{1}{e} \operatorname{Subst}\left[\int \frac{1}{a + b x} \, dx, x, \operatorname{Cot}\left[\frac{1}{2} (d + e x) + \frac{\pi}{4}\right]\right]$$

Program code:

```
Int[1/(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  Module[{f=FreeFactors[Cot[(d+e*x)/2+Pi/4],x]},
  -f/e*Subst[Int[1/(a+b*f*x),x],x,Cot[(d+e*x)/2+Pi/4]/f]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[a-c,0] && NeQ[a-b,0]
```

4: $\int \frac{1}{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]} \, dx$ when $a^2 - b^2 - c^2 \neq 0$

Reference: G&R 2.558.4

Derivation: Integration by substitution

- **Basis:** $F[\operatorname{Sin}[d + e x], \operatorname{Cos}[d + e x]] == \frac{2}{e} \operatorname{Subst}\left[\frac{1}{1+x^2} F\left[\frac{2 x}{1+x^2}, \frac{1-x^2}{1+x^2}\right], x, \operatorname{Tan}\left[\frac{1}{2} (d + e x)\right]\right] \partial_x \operatorname{Tan}\left[\frac{1}{2} (d + e x)\right]$

- **Basis:** $\frac{1}{a+b \operatorname{Cos}[d+e x]+c \operatorname{Sin}[d+e x]} == \frac{2}{e} \operatorname{Subst}\left[\frac{1}{a+b+2 c x+(a-b) x^2}, x, \operatorname{Tan}\left[\frac{1}{2} (d + e x)\right]\right] \partial_x \operatorname{Tan}\left[\frac{1}{2} (d + e x)\right]$

Rule: If $a^2 - b^2 - c^2 \neq 0$, **then**

$$\int \frac{1}{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]} \, dx \longrightarrow \frac{2}{e} \operatorname{Subst}\left[\int \frac{1}{a + b + 2 c x + (a - b) x^2} \, dx, x, \operatorname{Tan}\left[\frac{1}{2} (d + e x)\right]\right]$$

Program code:

```
Int[1/(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  Module[{f=FreeFactors[Tan[(d+e*x)/2],x]},
  2*f/e*Subst[Int[1/(a+b+2*c*f*x+(a-b)*f^2*x^2),x],x,Tan[(d+e*x)/2]/f]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[a^2-b^2-c^2,0]
```

2. $\int \frac{1}{\sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]}} \, dx$ when $a^2 - b^2 - c^2 \neq 0$

1: $\int \frac{1}{\sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]}} \, dx$ when $b^2 + c^2 == 0$

Reference: Integration by substitution

- Basis: If $b^2 + c^2 == 0$, then $f[b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]] == \frac{b f[b \operatorname{Cos}[d+e x]+c \operatorname{Sin}[d+e x]]}{c e (b \operatorname{Cos}[d+e x]+c \operatorname{Sin}[d+e x])} \partial_x (b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])$

- Rule: If $b^2 + c^2 == 0$, then

$$\int \frac{1}{\sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]}} \, dx \to \frac{b}{c e} \operatorname{Subst}\left[\int \frac{1}{x \sqrt{a + x}} \, dx, x, b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]\right]$$

- Program code:

```
Int[1/Sqrt[a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]],x_Symbol] :=
  b/(c*e)*Subst[Int[1/(x*Sqrt[a+x]),x],x,b*Cos[d+e*x]+c*Sin[d+e*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[b^2+c^2,0]
```

2. $\int \frac{1}{\sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]}} \, dx$ when $a^2 - b^2 - c^2 \neq 0 \wedge b^2 + c^2 \neq 0$

1: $\int \frac{1}{\sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]}} \, dx$ when $b^2 + c^2 \neq 0 \bigwedge a + \sqrt{b^2 + c^2} > 0$

- Derivation: Algebraic simplification

- Basis: If $b^2 + c^2 \neq 0$, then $a + b \operatorname{Cos}[z] + c \operatorname{Sin}[z] == a + \sqrt{b^2 + c^2} \operatorname{Cos}[z - \operatorname{ArcTan}[b, c]]$

- Rule: If $b^2 + c^2 \neq 0 \bigwedge a + \sqrt{b^2 + c^2} > 0$, then

$$\int \frac{1}{\sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]}} \, dx \to \int \frac{1}{\sqrt{a + \sqrt{b^2 + c^2} \operatorname{Cos}[d + e x - \operatorname{ArcTan}[b, c]]}} \, dx$$

- Program code:

```
Int[1/Sqrt[a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]],x_Symbol] :=
  Int[1/Sqrt[a+Sqrt[b^2+c^2]*Cos[d+e*x-ArcTan[b,c]]],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2+c^2,0] && GtQ[a+Sqrt[b^2+c^2],0]
```

2: $\int \frac{1}{\sqrt{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}}\,dx$ when $a^2 - b^2 - c^2 \neq 0 \bigwedge b^2 + c^2 \neq 0 \bigwedge \neg\left(a + \sqrt{b^2 + c^2} > 0\right)$

Derivation: Piecewise constant extraction and algebraic simplification

- **Basis:** $\partial_x \frac{\sqrt{\frac{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}{a+\sqrt{b^2+c^2}}}}{\sqrt{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}} == 0$
- **Basis: If** $b^2 + c^2 \neq 0$, **then** $a + b\,\mathrm{Cos}[z] + c\,\mathrm{Sin}[z] == a + \sqrt{b^2 + c^2}\,\mathrm{Cos}[z - \mathrm{ArcTan}[b,\ c]]$
- **Rule: If** $a^2 - b^2 - c^2 \neq 0 \bigwedge b^2 + c^2 \neq 0 \bigwedge \neg\left(a + \sqrt{b^2 + c^2} > 0\right)$, **then**

$$\int \frac{1}{\sqrt{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}}\,dx \longrightarrow$$

$$\frac{\sqrt{\frac{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}{a+\sqrt{b^2+c^2}}}}{\sqrt{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}} \int \frac{1}{\sqrt{\frac{a}{a+\sqrt{b^2+c^2}} + \frac{\sqrt{b^2+c^2}}{a+\sqrt{b^2+c^2}}\,\mathrm{Cos}[d + e\,x - \mathrm{ArcTan}[b,\ c]]}}\,dx$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]],x_Symbol] :=
  Sqrt[(a+b*Cos[d+e*x]+c*Sin[d+e*x])/(a+Sqrt[b^2+c^2])]/Sqrt[a+b*Cos[d+e*x]+c*Sin[d+e*x]]*
    Int[1/Sqrt[a/(a+Sqrt[b^2+c^2])+Sqrt[b^2+c^2]/(a+Sqrt[b^2+c^2])*Cos[d+e*x-ArcTan[b,c]]],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[a^2-b^2-c^2,0] && NeQ[b^2+c^2,0] && Not[GtQ[a+Sqrt[b^2+c^2],0]]
```

3. $\int (a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x])^n\,dx$ when $a^2 - b^2 - c^2 \neq 0 \bigwedge n < -1$

1: $\int \frac{1}{(a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x])^{3/2}}\,dx$ when $a^2 - b^2 - c^2 \neq 0$

- **Reference: G&R 2.558.1 with** $n = -\frac{3}{2}$
- **Rule: If** $a^2 - b^2 - c^2 \neq 0$, **then**

$$\int \frac{1}{(a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x])^{3/2}}\,dx \longrightarrow$$

$$\frac{2\,(c\,\mathrm{Cos}[d+e\,x]-b\,\mathrm{Sin}[d+e\,x])}{e\,\left(a^2-b^2-c^2\right)\sqrt{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}}+\frac{1}{a^2-b^2-c^2}\int\sqrt{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}\,dx$$

Program code:

```
Int[1/(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^(3/2),x_Symbol] :=
  2*(c*Cos[d+e*x]-b*Sin[d+e*x])/(e*(a^2-b^2-c^2)*Sqrt[a+b*Cos[d+e*x]+c*Sin[d+e*x]]) +
  1/(a^2-b^2-c^2)*Int[Sqrt[a+b*Cos[d+e*x]+c*Sin[d+e*x]],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[a^2-b^2-c^2,0]
```

2: $\int(a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n\,dx$ when $a^2-b^2-c^2\neq 0\bigwedge n<-1\bigwedge n\neq -\frac{3}{2}$

Reference: G&R 2.558.1

Rule: If $a^2-b^2-c^2\neq 0\bigwedge n<-1\bigwedge n\neq -\frac{3}{2}$**, then**

$$\int(a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n\,dx\longrightarrow$$

$$\frac{(-c\,\mathrm{Cos}[d+e\,x]+b\,\mathrm{Sin}[d+e\,x])\,(a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^{n+1}}{e\,(n+1)\,\left(a^2-b^2-c^2\right)}+$$

$$\frac{1}{(n+1)\,\left(a^2-b^2-c^2\right)}\int(a\,(n+1)-b\,(n+2)\,\mathrm{Cos}[d+e\,x]-c\,(n+2)\,\mathrm{Sin}[d+e\,x])\,(a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^{n+1}\,dx$$

Program code:

```
Int[(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_,x_Symbol] :=
  (-c*Cos[d+e*x]+b*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1)/(e*(n+1)*(a^2-b^2-c^2)) +
  1/((n+1)*(a^2-b^2-c^2))*
    Int[(a*(n+1)-b*(n+2)*Cos[d+e*x]-c*(n+2)*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[a^2-b^2-c^2,0] && LtQ[n,-1] && NeQ[n,-3/2]
```

2. $\int (A + B\,\mathrm{Cos}[d + e\,x] + C\,\mathrm{Sin}[d + e\,x])\,(a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x])^n\,dx$

1. $\int \frac{A + B\,\mathrm{Cos}[d + e\,x] + C\,\mathrm{Sin}[d + e\,x]}{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}\,dx$

1: $\int \frac{A + B\,\mathrm{Cos}[d + e\,x] + C\,\mathrm{Sin}[d + e\,x]}{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}\,dx$ when $b^2 + c^2 == 0$

Note: Although exactly analogous to G&R 2.451.3 for hyperbolic functions, there is no corresponding G&R 2.558.n formula for trig functions. Apparently the authors did not anticipate $b^2 + c^2$ could be 0 in the complex plane.

Rule: If $b^2 + c^2 == 0$, then

$$\int \frac{A + B\,\mathrm{Cos}[d + e\,x] + C\,\mathrm{Sin}[d + e\,x]}{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}\,dx \longrightarrow$$

$$\frac{(2\,a\,A - b\,B - c\,C)\,x}{2\,a^2} - \frac{(b\,B + c\,C)\,(b\,\mathrm{Cos}[d + e\,x] - c\,\mathrm{Sin}[d + e\,x])}{2\,a\,b\,c\,e} +$$
$$\frac{\left(a^2\,(b\,B - c\,C) - 2\,a\,A\,b^2 + b^2\,(b\,B + c\,C)\right)\,\mathrm{Log}[a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]]}{2\,a^2\,b\,c\,e}$$

Program code:

```
Int[(A_.+B_.*cos[d_.+e_.*x_]+C_.*sin[d_.+e_.*x_])/(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  (2*a*A-b*B-c*C)*x/(2*a^2) - (b*B+c*C)*(b*Cos[d+e*x]-c*Sin[d+e*x])/(2*a*b*c*e) +
  (a^2*(b*B-c*C)-2*a*A*b^2+b^2*(b*B+c*C))*Log[RemoveContent[a+b*Cos[d+e*x]+c*Sin[d+e*x],x]]/(2*a^2*b*c*e) /;
FreeQ[{a,b,c,d,e,A,B,C},x] && EqQ[b^2+c^2,0]
```

```
Int[(A_.+C_.*sin[d_.+e_.*x_])/(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  (2*a*A-c*C)*x/(2*a^2) - C*Cos[d+e*x]/(2*a*e) + c*C*Sin[d+e*x]/(2*a*b*e) +
  (-a^2*C+2*a*c*A+b^2*C)*Log[RemoveContent[a+b*Cos[d+e*x]+c*Sin[d+e*x],x]]/(2*a^2*b*e) /;
FreeQ[{a,b,c,d,e,A,C},x] && EqQ[b^2+c^2,0]
```

```
Int[(A_.+B_.*cos[d_.+e_.*x_])/(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  (2*a*A-b*B)*x/(2*a^2) - b*B*Cos[d+e*x]/(2*a*c*e) + B*Sin[d+e*x]/(2*a*e) +
  (a^2*B-2*a*b*A+b^2*B)*Log[RemoveContent[a+b*Cos[d+e*x]+c*Sin[d+e*x],x]]/(2*a^2*c*e) /;
FreeQ[{a,b,c,d,e,A,B},x] && EqQ[b^2+c^2,0]
```

2. $\int \frac{A + B\,\mathrm{Cos}[d + e\,x] + C\,\mathrm{Sin}[d + e\,x]}{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}\,dx$ when $b^2 + c^2 \neq 0$

1: $\int \frac{A + B \cos[d + e x] + C \sin[d + e x]}{a + b \cos[d + e x] + c \sin[d + e x]} \, dx$ when $b^2 + c^2 \neq 0 \land A \left(b^2 + c^2\right) - a (b B + c C) == 0$

- **Reference:** G&R 2.558.2 with $A \left(b^2 + c^2\right) - a (b B + c C) == 0$

- **Rule:** If $b^2 + c^2 \neq 0 \land A \left(b^2 + c^2\right) - a (b B + c C) == 0$, then

$$\int \frac{A + B \cos[d + e x] + C \sin[d + e x]}{a + b \cos[d + e x] + c \sin[d + e x]} \, dx \longrightarrow \frac{(b B + c C) x}{b^2 + c^2} + \frac{(c B - b C) \operatorname{Log}[a + b \cos[d + e x] + c \sin[d + e x]]}{e \left(b^2 + c^2\right)}$$

- **Program code:**

```
Int[(A_.+B_.*cos[d_.+e_.*x_]+C_.*sin[d_.+e_.*x_])/(a_.+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  (b*B+c*C)*x/(b^2+c^2) + (c*B-b*C)*Log[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(e*(b^2+c^2)) /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NeQ[b^2+c^2,0] && EqQ[A*(b^2+c^2)-a*(b*B+c*C),0]
```

```
Int[(A_.+C_.*sin[d_.+e_.*x_])/(a_.+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  c*C*x/(b^2+c^2) - b*C*Log[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(e*(b^2+c^2)) /;
FreeQ[{a,b,c,d,e,A,C},x] && NeQ[b^2+c^2,0] && EqQ[A*(b^2+c^2)-a*c*C,0]
```

```
Int[(A_.+B_.*cos[d_.+e_.*x_])/(a_.+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  b*B*x/(b^2+c^2) + c*B*Log[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(e*(b^2+c^2)) /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2+c^2,0] && EqQ[A*(b^2+c^2)-a*b*B,0]
```

2: $\int \frac{A+B\,\text{Cos}[d+e\,x]+C\,\text{Sin}[d+e\,x]}{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}\,dx$ when $b^2+c^2\neq 0\,\wedge\,A\left(b^2+c^2\right)-a\,(b\,B+c\,C)\neq 0$

- **Reference: G&R 2.558.2**

- **Rule: If** $b^2+c^2\neq 0\,\wedge\,A\left(b^2+c^2\right)-a\,(b\,B+c\,C)\neq 0$**, then**

$$\int \frac{A+B\,\text{Cos}[d+e\,x]+C\,\text{Sin}[d+e\,x]}{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}\,dx\,\longrightarrow$$

$$\frac{(b\,B+c\,C)\,x}{b^2+c^2}+\frac{(c\,B-b\,C)\,\text{Log}[a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]]}{e\left(b^2+c^2\right)}+$$

$$\frac{A\left(b^2+c^2\right)-a\,(b\,B+c\,C)}{b^2+c^2}\int\frac{1}{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}\,dx$$

Program code:

```
Int[(A_.+B_.*cos[d_.+e_.*x_]+C_.*sin[d_.+e_.*x_])/(a_.+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  (b*B+c*C)*x/(b^2+c^2) + (c*B-b*C)*Log[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(e*(b^2+c^2)) +
  (A*(b^2+c^2)-a*(b*B+c*C))/(b^2+c^2)*Int[1/(a+b*Cos[d+e*x]+c*Sin[d+e*x]),x] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NeQ[b^2+c^2,0] && NeQ[A*(b^2+c^2)-a*(b*B+c*C),0]
```

```
Int[(A_.+C_.*sin[d_.+e_.*x_])/(a_.+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  c*C*(d+e*x)/(e*(b^2+c^2)) - b*C*Log[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(e*(b^2+c^2)) +
  (A*(b^2+c^2)-a*c*C)/(b^2+c^2)*Int[1/(a+b*Cos[d+e*x]+c*Sin[d+e*x]),x] /;
FreeQ[{a,b,c,d,e,A,C},x] && NeQ[b^2+c^2,0] && NeQ[A*(b^2+c^2)-a*c*C,0]
```

```
Int[(A_.+B_.*cos[d_.+e_.*x_])/(a_.+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]),x_Symbol] :=
  b*B*(d+e*x)/(e*(b^2+c^2)) +
  c*B*Log[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(e*(b^2+c^2)) +
  (A*(b^2+c^2)-a*b*B)/(b^2+c^2)*Int[1/(a+b*Cos[d+e*x]+c*Sin[d+e*x]),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2+c^2,0] && NeQ[A*(b^2+c^2)-a*b*B,0]
```

2. $\int (A + B \operatorname{Cos}[d + e x] + C \operatorname{Sin}[d + e x]) (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx$ when $n \neq -1$

1. $\int (A + B \operatorname{Cos}[d + e x] + C \operatorname{Sin}[d + e x]) (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx$ when $n \neq -1 \land a^2 - b^2 - c^2 == 0$

1: $\int (A + B \operatorname{Cos}[d + e x] + C \operatorname{Sin}[d + e x]) (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx$ when $n \neq -1 \land a^2 - b^2 - c^2 == 0 \land (b B + c C) n + a A (n + 1) == 0$

- **Reference: G&R 2.558.1b**

- **Rule: If** $n \neq -1 \land a^2 - b^2 - c^2 == 0 \land (b B + c C) n + a A (n + 1) == 0$, **then**

$$\int (A + B \operatorname{Cos}[d + e x] + C \operatorname{Sin}[d + e x]) (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx \longrightarrow$$
$$\frac{1}{a e (n + 1)} (B c - b C - a C \operatorname{Cos}[d + e x] + a B \operatorname{Sin}[d + e x]) (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n$$

- **Program code:**

```
Int[(A_.+B_.*cos[d_.+e_.*x_]+C_.*sin[d_.+e_.*x_])*(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_.,x_Symbol] :=
  (B*c-b*C-a*C*Cos[d+e*x]+a*B*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n/(a*e*(n+1)) /;
FreeQ[{a,b,c,d,e,A,B,C,n},x] && NeQ[n,-1] && EqQ[a^2-b^2-c^2,0] && EqQ[(b*B+c*C)*n+a*A*(n+1),0]
```

```
Int[(A_.+C_.*sin[d_.+e_.*x_])*(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_.,x_Symbol] :=
  -(b*C+a*C*Cos[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n/(a*e*(n+1)) /;
FreeQ[{a,b,c,d,e,A,C,n},x] && NeQ[n,-1] && EqQ[a^2-b^2-c^2,0] && EqQ[c*C*n+a*A*(n+1),0]
```

```
Int[(A_.+B_.*cos[d_.+e_.*x_])*(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_.,x_Symbol] :=
  (B*c+a*B*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n/(a*e*(n+1)) /;
FreeQ[{a,b,c,d,e,A,B,n},x] && NeQ[n,-1] && EqQ[a^2-b^2-c^2,0] && EqQ[b*B*n+a*A*(n+1),0]
```

2: $\int (A + B \operatorname{Cos}[d + e x] + C \operatorname{Sin}[d + e x]) (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx$ when $n \neq -1 \land a^2 - b^2 - c^2 == 0 \land (b B + c C) n + a A (n + 1) \neq 0$

- **Reference: G&R 2.558.1b**

- **Rule: If** $n \neq -1 \land a^2 - b^2 - c^2 == 0 \land (b B + c C) n + a A (n + 1) \neq 0$, **then**

$$\int (A + B \operatorname{Cos}[d + e x] + C \operatorname{Sin}[d + e x]) (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx \longrightarrow$$
$$\frac{1}{a e (n + 1)} (B c - b C - a C \operatorname{Cos}[d + e x] + a B \operatorname{Sin}[d + e x]) (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n +$$

$$\frac{(b\,B+c\,C)\,n+a\,A\,(n+1)}{a\,(n+1)}\int(a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x])^n\,dx$$

Program code:

```
Int[(A_.+B_.*cos[d_.+e_.*x_]+C_.*sin[d_.+e_.*x_])*(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_.,x_Symbol] :=
  (B*c-b*C-a*C*Cos[d+e*x]+a*B*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n/(a*e*(n+1)) +
  ((b*B+c*C)*n+a*A*(n+1))/(a*(n+1))*Int[(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n,x] /;
FreeQ[{a,b,c,d,e,A,B,C,n},x] && NeQ[n,-1] && EqQ[a^2-b^2-c^2,0] && NeQ[(b*B+c*C)*n+a*A*(n+1),0]
```

```
Int[(A_.+C_.*sin[d_.+e_.*x_])*(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_.,x_Symbol] :=
  -(b*C+a*C*Cos[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n/(a*e*(n+1)) +
  (c*C*n+a*A*(n+1))/(a*(n+1))*Int[(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n,x] /;
FreeQ[{a,b,c,d,e,A,C,n},x] && NeQ[n,-1] && EqQ[a^2-b^2-c^2,0] && NeQ[c*C*n+a*A*(n+1),0]
```

```
Int[(A_.+B_.*cos[d_.+e_.*x_])*(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_.,x_Symbol] :=
  (B*c+a*B*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n/(a*e*(n+1)) +
  (b*B*n+a*A*(n+1))/(a*(n+1))*Int[(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n,x] /;
FreeQ[{a,b,c,d,e,A,B,n},x] && NeQ[n,-1] && EqQ[a^2-b^2-c^2,0] && NeQ[b*B*n+a*A*(n+1),0]
```

2. $\int(A+B\,\text{Cos}[d+e\,x]+C\,\text{Sin}[d+e\,x])\,(a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x])^n\,dx$ when $n\neq-1\,\wedge\,a^2-b^2-c^2\neq0$

1: $\int(B\,\text{Cos}[d+e\,x]+C\,\text{Sin}[d+e\,x])\,(b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x])^n\,dx$ when $n\neq-1\,\wedge\,b^2+c^2\neq0\,\wedge\,b\,B+c\,C==0$

Reference: G&R 2.558.1a with $a=0$, $A=0$ and $b\,B+c\,C==0$

Rule: If $n\neq-1\,\wedge\,b^2+c^2\neq0\,\wedge\,b\,B+c\,C==0$, then

$$\int(B\,\text{Cos}[d+e\,x]+C\,\text{Sin}[d+e\,x])\,(b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x])^n\,dx\longrightarrow\frac{(c\,B-b\,C)\,(b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x])^{n+1}}{e\,(n+1)\,\left(b^2+c^2\right)}$$

Program code:

```
Int[(B_.*cos[d_.+e_.*x_]+C_.*sin[d_.+e_.*x_])*(b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_.,x_Symbol] :=
  (c*B-b*C)*(b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1)/(e*(n+1)*(b^2+c^2)) /;
FreeQ[{b,c,d,e,B,C},x] && NeQ[n,-1] && NeQ[b^2+c^2,0] && EqQ[b*B+c*C,0]
```

2: $\int (A + B\,\text{Cos}[d + e\,x] + C\,\text{Sin}[d + e\,x])\,(a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])^n\,dx$ when $n > 0\,\wedge\,a^2 - b^2 - c^2 \neq 0$

Reference: G&R 2.558.1a inverted

Rule: If $n > 0\,\wedge\,a^2 - b^2 - c^2 \neq 0$, then

$$\int (A + B\,\text{Cos}[d + e\,x] + C\,\text{Sin}[d + e\,x])\,(a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])^n\,dx \longrightarrow$$

$$\frac{1}{a\,e\,(n+1)}\,(B\,c - b\,C - a\,C\,\text{Cos}[d + e\,x] + a\,B\,\text{Sin}[d + e\,x])\,(a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])^n +$$

$$\frac{1}{a\,(n+1)}\int (a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])^{n-1}\cdot$$

$$\left(a\,(b\,B + c\,C)\,n + a^2\,A\,(n+1) + \left(n\left(a^2\,B - B\,c^2 + b\,c\,C\right) + a\,b\,A\,(n+1)\right)\text{Cos}[d + e\,x] + \left(n\left(b\,B\,c + a^2\,C - b^2\,C\right) + a\,c\,A\,(n+1)\right)\text{Sin}[d + e\,x]\right)dx$$

Program code:

```
Int[(A_.+B_.*cos[d_.+e_.*x_]+C_.*sin[d_.+e_.*x_])*(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_.,x_Symbol] :=
  (B*c-b*C-a*C*Cos[d+e*x]+a*B*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n/(a*e*(n+1)) +
  1/(a*(n+1))*Int[(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-1)*
    Simp[a*(b*B+c*C)*n+a^2*A*(n+1)+
      (n*(a^2*B-B*c^2+b*c*C)+a*b*A*(n+1))*Cos[d+e*x]+
      (n*(b*B*c+a^2*C-b^2*C)+a*c*A*(n+1))*Sin[d+e*x],x],x] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && GtQ[n,0] && NeQ[a^2-b^2-c^2,0]
```

```
Int[(A_.+C_.*sin[d_.+e_.*x_])*(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_.,x_Symbol] :=
  -(b*C+a*C*Cos[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n/(a*e*(n+1)) +
  1/(a*(n+1))*Int[(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-1)*
    Simp[a*c*C*n+a^2*A*(n+1)+(c*b*C*n+a*b*A*(n+1))*Cos[d+e*x]+(a^2*C*n-b^2*C*n+a*c*A*(n+1))*Sin[d+e*x],x],x] /;
FreeQ[{a,b,c,d,e,A,C},x] && GtQ[n,0] && NeQ[a^2-b^2-c^2,0]
```

```
Int[(A_.+B_.*cos[d_.+e_.*x_])*(a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_.,x_Symbol] :=
  (B*c+a*B*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n/(a*e*(n+1)) +
  1/(a*(n+1))*Int[(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-1)*
    Simp[a*b*B*n+a^2*A*(n+1)+(a^2*B*n-c^2*B*n+a*b*A*(n+1))*Cos[d+e*x]+(b*c*B*n+a*c*A*(n+1))*Sin[d+e*x],x],x] /;
FreeQ[{a,b,c,d,e,A,B},x] && GtQ[n,0] && NeQ[a^2-b^2-c^2,0]
```

3. $\int (A + B\,\text{Cos}[d + e\,x] + C\,\text{Sin}[d + e\,x])\,(a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])^n\,dx$ when $n < 0\,\wedge\,a^2 - b^2 - c^2 \neq 0\,\wedge\,n \neq -1$

1: $\int \frac{A + B \operatorname{Cos}[d + e x] + C \operatorname{Sin}[d + e x]}{\sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]}} \, dx$ when $B c - b C == 0 \wedge A b - a B \neq 0$

Derivation: Algebraic simplification

Basis: If $B c - b C == 0$, **then** $A + B z + C w == \frac{B}{b} (a + b z + c w) + \frac{A b - a B}{b}$

Rule: If $B c - b C == 0 \wedge A b - a B \neq 0$, **then**

$$\int \frac{A + B \operatorname{Cos}[d + e x] + C \operatorname{Sin}[d + e x]}{\sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]}} \, dx \longrightarrow \frac{B}{b} \int \sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]} \, dx + \frac{A b - a B}{b} \int \frac{1}{\sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]}} \, dx$$

Program code:

```
Int[(A_.+B_.*cos[d_.+e_.*x_]+C_.*sin[d_.+e_.*x_])/Sqrt[a_+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]],x_Symbol] :=
  B/b*Int[Sqrt[a+b*Cos[d+e*x]+c*Sin[d+e*x]],x] +
  (A*b-a*B)/b*Int[1/Sqrt[a+b*Cos[d+e*x]+c*Sin[d+e*x]],x] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && EqQ[B*c-b*C,0] && NeQ[A*b-a*B,0]
```

2. $\int (A + B \operatorname{Cos}[d + e x] + C \operatorname{Sin}[d + e x]) (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx$ when $n < -1 \wedge a^2 - b^2 - c^2 \neq 0$

1. $\int \frac{A + B \operatorname{Cos}[d + e x] + C \operatorname{Sin}[d + e x]}{(a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^2} \, dx$ when $a^2 - b^2 - c^2 \neq 0$

1: $\int \frac{A + B \operatorname{Cos}[d + e x] + C \operatorname{Sin}[d + e x]}{(a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^2} \, dx$ when $a^2 - b^2 - c^2 \neq 0 \wedge a A - b B - c C == 0$

Reference: G&R 2.558.1a with $n = -2$ **and** $a A - b B - c C = 0$

Rule: If $a^2 - b^2 - c^2 \neq 0 \wedge a A - b B - c C == 0$, **then**

$$\int \frac{A + B \operatorname{Cos}[d + e x] + C \operatorname{Sin}[d + e x]}{(a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^2} \, dx \longrightarrow \frac{c B - b C - (a C - c A) \operatorname{Cos}[d + e x] + (a B - b A) \operatorname{Sin}[d + e x]}{e \left(a^2 - b^2 - c^2\right) (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])}$$

Program code:

```
Int[(A_.+B_.*cos[d_.+e_.*x_]+C_.*sin[d_.+e_.*x_])/(a_.+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^2,x_Symbol] :=
  (c*B-b*C-(a*C-c*A)*Cos[d+e*x]+(a*B-b*A)*Sin[d+e*x])/
    (e*(a^2-b^2-c^2)*(a+b*Cos[d+e*x]+c*Sin[d+e*x])) /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NeQ[a^2-b^2-c^2,0] && EqQ[a*A-b*B-c*C,0]
```

```
Int[(A_.+C_.*sin[d_.+e_.*x_])/(a_.+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^2,x_Symbol] :=
  -(b*C+(a*C-c*A)*Cos[d+e*x]+b*A*Sin[d+e*x])/(e*(a^2-b^2-c^2)*(a+b*Cos[d+e*x]+c*Sin[d+e*x])) /;
FreeQ[{a,b,c,d,e,A,C},x] && NeQ[a^2-b^2-c^2,0] && EqQ[a*A-c*C,0]
```

```
Int[(A_.+B_.*cos[d_.+e_.*x_])/(a_.+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^2,x_Symbol] :=
  (c*B+c*A*Cos[d+e*x]+(a*B-b*A)*Sin[d+e*x])/(e*(a^2-b^2-c^2)*(a+b*Cos[d+e*x]+c*Sin[d+e*x])) /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[a^2-b^2-c^2,0] && EqQ[a*A-b*B,0]
```

2: $\int \frac{A + B\,\text{Cos}[d + e\,x] + C\,\text{Sin}[d + e\,x]}{(a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])^2}\,dx$ when $a^2 - b^2 - c^2 \neq 0 \land a\,A - b\,B - c\,C \neq 0$

Reference: G&R 2.558.1a with $n = -2$

–

Rule: If $a^2 - b^2 - c^2 \neq 0 \land a\,A - b\,B - c\,C \neq 0$, then

$$\int \frac{A + B\,\text{Cos}[d + e\,x] + C\,\text{Sin}[d + e\,x]}{(a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])^2}\,dx \longrightarrow$$

$$\frac{c\,B - b\,C - (a\,C - c\,A)\,\text{Cos}[d + e\,x] + (a\,B - b\,A)\,\text{Sin}[d + e\,x]}{e\left(a^2 - b^2 - c^2\right)(a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])} + \frac{a\,A - b\,B - c\,C}{a^2 - b^2 - c^2}\int \frac{1}{a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x]}\,dx$$

Program code:

```
Int[(A_.+B_.*cos[d_.+e_.*x_]+C_.*sin[d_.+e_.*x_])/(a_.+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^2,x_Symbol] :=
  (c*B-b*C-(a*C-c*A)*Cos[d+e*x]+(a*B-b*A)*Sin[d+e*x])/
    (e*(a^2-b^2-c^2)*(a+b*Cos[d+e*x]+c*Sin[d+e*x])) +
  (a*A-b*B-c*C)/(a^2-b^2-c^2)*Int[1/(a+b*Cos[d+e*x]+c*Sin[d+e*x]),x] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NeQ[a^2-b^2-c^2,0] && NeQ[a*A-b*B-c*C,0]
```

```
Int[(A_.+C_.*sin[d_.+e_.*x_])/(a_.+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^2,x_Symbol] :=
  -(b*C+(a*C-c*A)*Cos[d+e*x]+b*A*Sin[d+e*x])/(e*(a^2-b^2-c^2)*(a+b*Cos[d+e*x]+c*Sin[d+e*x])) +
  (a*A-c*C)/(a^2-b^2-c^2)*Int[1/(a+b*Cos[d+e*x]+c*Sin[d+e*x]),x] /;
FreeQ[{a,b,c,d,e,A,C},x] && NeQ[a^2-b^2-c^2,0] && NeQ[a*A-c*C,0]
```

```
Int[(A_.+B_.*cos[d_.+e_.*x_])/(a_.+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^2,x_Symbol] :=
  (c*B+c*A*Cos[d+e*x]+(a*B-b*A)*Sin[d+e*x])/(e*(a^2-b^2-c^2)*(a+b*Cos[d+e*x]+c*Sin[d+e*x])) +
  (a*A-b*B)/(a^2-b^2-c^2)*Int[1/(a+b*Cos[d+e*x]+c*Sin[d+e*x]),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[a^2-b^2-c^2,0] && NeQ[a*A-b*B,0]
```

2: $\int (A+B\cos[d+e x]+C\sin[d+e x])\,(a+b\cos[d+e x]+c\sin[d+e x])^n\,dx$ when $n<-1\wedge a^2-b^2-c^2\neq 0\wedge n\neq -2$

Reference: G&R 2.558.1a

Rule: If $n<-1\wedge a^2-b^2-c^2\neq 0\wedge n\neq -2$, then

$$\int (A+B\cos[d+e x]+C\sin[d+e x])\,(a+b\cos[d+e x]+c\sin[d+e x])^n\,dx\longrightarrow$$
$$-\left((c B-b C-(a C-c A)\cos[d+e x]+(a B-b A)\sin[d+e x])\,(a+b\cos[d+e x]+c\sin[d+e x])^{n+1}\right)\Big/\left(e\,(n+1)\left(a^2-b^2-c^2\right)\right)+$$
$$\frac{1}{(n+1)\left(a^2-b^2-c^2\right)}\int (a+b\cos[d+e x]+c\sin[d+e x])^{n+1}\cdot$$
$$((n+1)\,(a A-b B-c C)+(n+2)\,(a B-b A)\cos[d+e x]+(n+2)\,(a C-c A)\sin[d+e x])\,dx$$

Program code:

```
Int[(A_.+B_.*cos[d_.+e_.*x_]+C_.*sin[d_.+e_.*x_])*(a_.+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_,x_Symbol] :=
  -(c*B-b*C-(a*C-c*A)*Cos[d+e*x]+(a*B-b*A)*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1)/
    (e*(n+1)*(a^2-b^2-c^2)) +
  1/((n+1)*(a^2-b^2-c^2))*Int[(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1)*
    Simp[(n+1)*(a*A-b*B-c*C)+(n+2)*(a*B-b*A)*Cos[d+e*x]+(n+2)*(a*C-c*A)*Sin[d+e*x],x],x] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && LtQ[n,-1] && NeQ[a^2-b^2-c^2,0] && NeQ[n,-2]
```

```
Int[(A_.+C_.*sin[d_.+e_.*x_])*(a_.+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_,x_Symbol] :=
  (b*C+(a*C-c*A)*Cos[d+e*x]+b*A*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1)/
    (e*(n+1)*(a^2-b^2-c^2)) +
  1/((n+1)*(a^2-b^2-c^2))*Int[(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1)*
    Simp[(n+1)*(a*A-c*C)-(n+2)*b*A*Cos[d+e*x]+(n+2)*(a*C-c*A)*Sin[d+e*x],x],x] /;
FreeQ[{a,b,c,d,e,A,C},x] && LtQ[n,-1] && NeQ[a^2-b^2-c^2,0] && NeQ[n,-2]
```

```
Int[(A_.+B_.*cos[d_.+e_.*x_])*(a_.+b_.*cos[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_])^n_,x_Symbol] :=
  -(c*B+c*A*Cos[d+e*x]+(a*B-b*A)*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1)/
    (e*(n+1)*(a^2-b^2-c^2)) +
  1/((n+1)*(a^2-b^2-c^2))*Int[(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1)*
    Simp[(n+1)*(a*A-b*B)+(n+2)*(a*B-b*A)*Cos[d+e*x]-(n+2)*c*A*Sin[d+e*x],x],x] /;
FreeQ[{a,b,c,d,e,A,B},x] && LtQ[n,-1] && NeQ[a^2-b^2-c^2,0] && NeQ[n,-2]
```

3. $\int u\,(a+b\sec[d+e x]+c\tan[d+e x])^n\,dx$

1: $\int \frac{1}{a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x]}\,dx$

Derivation: Algebraic simplification

Rule:

$$\int \frac{1}{a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x]}\,dx \longrightarrow \int \frac{\mathrm{Cos}[d+e\,x]}{b+a\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}\,dx$$

Program code:

```
Int[1/(a_.+b_.*sec[d_.+e_.*x_]+c_.*tan[d_.+e_.*x_]),x_Symbol] :=
  Int[Cos[d+e*x]/(b+a*Cos[d+e*x]+c*Sin[d+e*x]),x] /;
FreeQ[{a,b,c,d,e},x]
```

```
Int[1/(a_.+b_.*csc[d_.+e_.*x_]+c_.*cot[d_.+e_.*x_]),x_Symbol] :=
  Int[Sin[d+e*x]/(b+a*Sin[d+e*x]+c*Cos[d+e*x]),x] /;
FreeQ[{a,b,c,d,e},x]
```

2. $\int \mathrm{Cos}[d+e\,x]^n\,(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x])^n\,dx$

1: $\int \mathrm{Cos}[d+e\,x]^n\,(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x])^n\,dx$ when $n\in\mathbb{Z}$

Derivation: Algebraic simplification

Rule: If $n\in\mathbb{Z}$**, then**

$$\int \mathrm{Cos}[d+e\,x]^n\,(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x])^n\,dx \longrightarrow \int (b+a\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n\,dx$$

Program code:

```
Int[cos[d_.+e_.*x_]^n_.*(a_.+b_.*sec[d_.+e_.*x_]+c_.*tan[d_.+e_.*x_])^n_.,x_Symbol] :=
  Int[(b+a*Cos[d+e*x]+c*Sin[d+e*x])^n,x] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[n]
```

```
Int[sin[d_.+e_.*x_]^n_.*(a_.+b_.*csc[d_.+e_.*x_]+c_.*cot[d_.+e_.*x_])^n_.,x_Symbol] :=
  Int[(b+a*Sin[d+e*x]+c*Cos[d+e*x])^n,x] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[n]
```

2: $\int \mathrm{Cos}[d+e\,x]^n\,(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x])^n\,dx$ when $n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\mathrm{Cos}[d+e\,x]^n\,(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x])^n}{(b+a\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n}==0$

Rule: If $n\in\mathbb{Z}$**, then**

$$\int \mathrm{Cos}[d+e\,x]^n\,(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x])^n\,dx \longrightarrow \frac{\mathrm{Cos}[d+e\,x]^n\,(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x])^n}{(b+a\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n}\int (b+a\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n\,dx$$

Program code:

```
Int[cos[d_.+e_.*x_]^n_*(a_.+b_.*sec[d_.+e_.*x_]+c_.*tan[d_.+e_.*x_])^n_,x_Symbol] :=
  Cos[d+e*x]^n*(a+b*Sec[d+e*x]+c*Tan[d+e*x])^n/(b+a*Cos[d+e*x]+c*Sin[d+e*x])^n*Int[(b+a*Cos[d+e*x]+c*Sin[d+e*x])^n,x] /;
FreeQ[{a,b,c,d,e},x] && Not[IntegerQ[n]]
```

```
Int[sin[d_.+e_.*x_]^n_*(a_.+b_.*csc[d_.+e_.*x_]+c_.*cot[d_.+e_.*x_])^n_,x_Symbol] :=
  Sin[d+e*x]^n*(a+b*Csc[d+e*x]+c*Cot[d+e*x])^n/(b+a*Sin[d+e*x]+c*Cos[d+e*x])^n*Int[(b+a*Sin[d+e*x]+c*Cos[d+e*x])^n,x] /;
FreeQ[{a,b,c,d,e},x] && Not[IntegerQ[n]]
```

3. $\int \frac{\mathrm{Sec}[d+e\,x]^n}{(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x])^n}\,dx$

1: $\int \frac{\mathrm{Sec}[d+e\,x]^n}{(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x])^n}\,dx$ when $n\in\mathbb{Z}$

Derivation: Algebraic simplification

Rule: If $n\in\mathbb{Z}$**, then**

$$\int \frac{\mathrm{Sec}[d+e\,x]^n}{(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x])^n}\,dx \longrightarrow \int \frac{1}{(b+a\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n}\,dx$$

Program code:

```
Int[sec[d_.+e_.*x_]^n_.*(a_.+b_.*sec[d_.+e_.*x_]+c_.*tan[d_.+e_.*x_])^m_,x_Symbol] :=
  Int[1/(b+a*Cos[d+e*x]+c*Sin[d+e*x])^n,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[m+n,0] && IntegerQ[n]
```

```
Int[csc[d_.+e_.*x_]^n_.*(a_.+b_.*csc[d_.+e_.*x_]+c_.*cot[d_.+e_.*x_])^m_,x_Symbol] :=
  Int[1/(b+a*Sin[d+e*x]+c*Cos[d+e*x])^n,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[m+n,0] && IntegerQ[n]
```

2: $\int \mathrm{Cos}[d+e\,x]^n\,(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x])^n\,dx$ when $n\notin\mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \frac{\mathrm{Sec}[d+e\,x]^n\,(b+a\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n}{(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x])^n} == 0$
- **Rule: If** $n\in\mathbb{Z}$**, then**

$$\int \frac{\mathrm{Sec}[d+e\,x]^n}{(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x])^n}\,dx \longrightarrow \frac{\mathrm{Sec}[d+e\,x]^n\,(b+a\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n}{(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Tan}[d+e\,x])^n}\int \frac{1}{(b+a\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n}\,dx$$

- **Program code:**

```
Int[sec[d_.+e_.*x_]^n_.*(a_.+b_.*sec[d_.+e_.*x_]+c_.*tan[d_.+e_.*x_])^m_,x_Symbol] :=
  Sec[d+e*x]^n*(b+a*Cos[d+e*x]+c*Sin[d+e*x])^n/(a+b*Sec[d+e*x]+c*Tan[d+e*x])^n*Int[1/(b+a*Cos[d+e*x]+c*Sin[d+e*x])^n,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[m+n,0] && Not[IntegerQ[n]]
```

```
Int[csc[d_.+e_.*x_]^n_.*(a_.+b_.*csc[d_.+e_.*x_]+c_.*cot[d_.+e_.*x_])^m_,x_Symbol] :=
  Csc[d+e*x]^n*(b+a*Sin[d+e*x]+c*Cos[d+e*x])^n/(a+b*Csc[d+e*x]+c*Cot[d+e*x])^n*Int[1/(b+a*Sin[d+e*x]+c*Cos[d+e*x])^n,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[m+n,0] && Not[IntegerQ[n]]
```