Rules for integrands of the form $(c+d\,x)^m \left(F^{g\,(e+f\,x)}\right)^n \left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p$

1. $\int (c+d\,x)^m \left(F^{g\,(e+f\,x)}\right)^n \left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p \, dx$

   1: $\int \frac{(c+d\,x)^m \left(F^{g\,(e+f\,x)}\right)^n}{a+b\left(F^{g\,(e+f\,x)}\right)^n} \, dx$ when $m \in \mathbb{Z}^+$

- **Derivation: Integration by parts**
- **Basis:** $\frac{\left(F^{g\,(e+f\,x)}\right)^n}{a+b\left(F^{g\,(e+f\,x)}\right)^n} == \partial_x \frac{\mathrm{Log}\left[1+\frac{b\left(F^{g\,(e+f\,x)}\right)^n}{a}\right]}{b\,f\,g\,n\,\mathrm{Log}[F]}$
- **Rule: If** $m \in \mathbb{Z}^+$**, then**

$$\int \frac{(c+d\,x)^m \left(F^{g\,(e+f\,x)}\right)^n}{a+b\left(F^{g\,(e+f\,x)}\right)^n} \, dx \longrightarrow \frac{(c+d\,x)^m}{b\,f\,g\,n\,\mathrm{Log}[F]} \mathrm{Log}\left[1+\frac{b\left(F^{g\,(e+f\,x)}\right)^n}{a}\right] - \frac{d\,m}{b\,f\,g\,n\,\mathrm{Log}[F]} \int (c+d\,x)^{m-1} \mathrm{Log}\left[1+\frac{b\left(F^{g\,(e+f\,x)}\right)^n}{a}\right] dx$$

- **Program code:**

```
Int[(c_.+d_.*x_)^m_.*(F_^(g_.*(e_.+f_.*x_)))^n_./(a_+b_.*(F_^(g_.*(e_.+f_.*x_)))^n_.),x_Symbol] :=
  (c+d*x)^m/(b*f*g*n*Log[F])*Log[1+b*(F^(g*(e+f*x)))^n/a] -
  d*m/(b*f*g*n*Log[F])*Int[(c+d*x)^(m-1)*Log[1+b*(F^(g*(e+f*x)))^n/a],x] /;
FreeQ[{F,a,b,c,d,e,f,g,n},x] && IGtQ[m,0]
```

2: $\int (c+d\,x)^m \left(F^{g\,(e+f\,x)}\right)^n \left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p dx$ when $p\neq -1$

- **Derivation: Integration by parts**
- **Basis:** $\left(F^{g\,(e+f\,x)}\right)^n \left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p == \partial_x \frac{\left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^{p+1}}{b\,f\,g\,n\,(p+1)\,\text{Log}[F]}$
- **Rule: If** $p\neq -1$**, then**

$$\int (c+d\,x)^m \left(F^{g\,(e+f\,x)}\right)^n \left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p dx \longrightarrow$$
$$\frac{(c+d\,x)^m \left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^{p+1}}{b\,f\,g\,n\,(p+1)\,\text{Log}[F]} - \frac{d\,m}{b\,f\,g\,n\,(p+1)\,\text{Log}[F]} \int (c+d\,x)^{m-1} \left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^{p+1} dx$$

- **Program code:**

```
Int[(c_.+d_.*x_)^m_.*(F_^(g_.*(e_.+f_.*x_)))^n_.*(a_.+b_.*(F_^(g_.*(e_.+f_.*x_)))^n_.)^p_.,x_Symbol] :=
  (c+d*x)^m*(a+b*(F^(g*(e+f*x)))^n)^(p+1)/(b*f*g*n*(p+1)*Log[F]) -
  d*m/(b*f*g*n*(p+1)*Log[F])*Int[(c+d*x)^(m-1)*(a+b*(F^(g*(e+f*x)))^n)^(p+1),x] /;
FreeQ[{F,a,b,c,d,e,f,g,m,n,p},x] && NeQ[p,-1]
```

X: $\int (c+d\,x)^m \left(F^{g\,(e+f\,x)}\right)^n \left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p dx$

- **Rule:**

$$\int (c+d\,x)^m \left(F^{g\,(e+f\,x)}\right)^n \left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p dx \longrightarrow \int (c+d\,x)^m \left(F^{g\,(e+f\,x)}\right)^n \left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p dx$$

- **Program code:**

```
Int[(c_.+d_.*x_)^m_.*(F_^(g_.*(e_.+f_.*x_)))^n_.*(a_.+b_.*(F_^(g_.*(e_.+f_.*x_)))^n_.)^p_.,x_Symbol] :=
  Unintegrable[(c+d*x)^m*(F^(g*(e+f*x)))^n*(a+b*(F^(g*(e+f*x)))^n)^p,x] /;
FreeQ[{F,a,b,c,d,e,f,g,m,n,p},x]
```

2: $\int (c+d x)^m \left(k G^{j (h+i x)}\right)^q \left(a+b \left(F^{g (e+f x)}\right)^n\right)^p dx$ when $f g n \operatorname{Log}[F]-i j q \operatorname{Log}[G]==0$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $f g n \operatorname{Log}[F]-i j q \operatorname{Log}[G]==0$**, then** $\partial_x \frac{\left(k G^{j (h+i x)}\right)^q}{\left(F^{g (e+f x)}\right)^n}==0$
- **Rule: If** $f g n \operatorname{Log}[F]-i j q \operatorname{Log}[G]==0$**, then**

$$\int (c+d x)^m \left(k G^{j (h+i x)}\right)^q \left(a+b \left(F^{g (e+f x)}\right)^n\right)^p dx \longrightarrow \frac{\left(k G^{j (h+i x)}\right)^q}{\left(F^{g (e+f x)}\right)^n} \int (c+d x)^m \left(F^{g (e+f x)}\right)^n \left(a+b \left(F^{g (e+f x)}\right)^n\right)^p dx$$

- **Program code:**

```
Int[(c_.+d_.*x_)^m_.*(k_.*G_^(j_.*(h_.+i_.*x_)))^q_.*(a_.+b_.*(F_^(g_.*(e_.+f_.*x_)))^n_.)^p_.,x_Symbol] :=
  (k*G^(j*(h+i*x)))^q/(F^(g*(e+f*x)))^n*Int[(c+d*x)^m*(F^(g*(e+f*x)))^n*(a+b*(F^(g*(e+f*x)))^n)^p,x] /;
FreeQ[{F,a,b,c,d,e,f,g,h,i,j,k,m,n,p,q},x] && EqQ[f*g*n*Log[F]-i*j*q*Log[G],0] && NeQ[(k*G^(j*(h+i*x)))^q-(F^(g*(e+f*x)))^n,0]
```