Rules for integrands of the form $(a + b \operatorname{ArcSin}[c x])^n$

1: $\int (a + b \operatorname{ArcSin}[c x])^n \, dx$ when $n > 0$

- **Derivation: Integration by parts**
- **Basis:** $\partial_x (a + b \operatorname{ArcSin}[c x])^n == \frac{b c n (a+b \operatorname{ArcSin}[c x])^{n-1}}{\sqrt{1-c^2 x^2}}$
- **Rule: If $n > 0$, then**

$$\int (a + b \operatorname{ArcSin}[c x])^n \, dx \longrightarrow x (a + b \operatorname{ArcSin}[c x])^n - b c n \int \frac{x (a + b \operatorname{ArcSin}[c x])^{n-1}}{\sqrt{1 - c^2 x^2}} \, dx$$

- **Program code:**

```
Int[(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  x*(a+b*ArcSin[c*x])^n -
  b*c*n*Int[x*(a+b*ArcSin[c*x])^(n-1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c},x] && GtQ[n,0]
```

```
Int[(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  x*(a+b*ArcCos[c*x])^n +
  b*c*n*Int[x*(a+b*ArcCos[c*x])^(n-1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c},x] && GtQ[n,0]
```

2: $\int (a + b\,\text{ArcSin}[c\,x])^n\,dx$ when $n < -1$

- **Derivation: Integration by parts**
- **Basis:** $\frac{(a+b\,\text{ArcSin}[c\,x])^n}{\sqrt{1-c^2\,x^2}} == \partial_x \frac{(a+b\,\text{ArcSin}[c\,x])^{n+1}}{b\,c\,(n+1)}$
- **Basis:** $\partial_x \sqrt{1-c^2\,x^2} == -\frac{c^2\,x}{\sqrt{1-c^2\,x^2}}$
- **Rule: If** $n < -1$**, then**

$$\int (a + b\,\text{ArcSin}[c\,x])^n\,dx \longrightarrow \frac{\sqrt{1-c^2\,x^2}\,(a+b\,\text{ArcSin}[c\,x])^{n+1}}{b\,c\,(n+1)} + \frac{c}{b\,(n+1)} \int \frac{x\,(a+b\,\text{ArcSin}[c\,x])^{n+1}}{\sqrt{1-c^2\,x^2}}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  Sqrt[1-c^2*x^2]*(a+b*ArcSin[c*x])^(n+1)/(b*c*(n+1)) +
  c/(b*(n+1))*Int[x*(a+b*ArcSin[c*x])^(n+1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c},x] && LtQ[n,-1]
```

```
Int[(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  -Sqrt[1-c^2*x^2]*(a+b*ArcCos[c*x])^(n+1)/(b*c*(n+1)) -
  c/(b*(n+1))*Int[x*(a+b*ArcCos[c*x])^(n+1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c},x] && LtQ[n,-1]
```

3: $\int (a+b\,\text{ArcSin}[c\,x])^n\,dx$

- **Derivation: Integration by substitution**
- **Basis:** $F[a+b\,\text{ArcSin}[c\,x]] == \frac{1}{b\,c}\,\text{Subst}\left[F[x]\,\text{Cos}\left[\frac{a}{b}-\frac{x}{b}\right],\,x,\,a+b\,\text{ArcSin}[c\,x]\right]\,\partial_x (a+b\,\text{ArcSin}[c\,x])$
- **Rule:**

$$\int (a+b\,\text{ArcSin}[c\,x])^n\,dx \longrightarrow \frac{1}{b\,c}\,\text{Subst}\left[\int x^n\,\text{Cos}\left[\frac{a}{b}-\frac{x}{b}\right]dx,\,x,\,a+b\,\text{ArcSin}[c\,x]\right]$$

- **Program code:**

```
Int[(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  1/(b*c)*Subst[Int[x^n*Cos[a/b-x/b],x],x,a+b*ArcSin[c*x]] /;
FreeQ[{a,b,c,n},x]
```

```
Int[(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  1/(b*c)*Subst[Int[x^n*Sin[a/b-x/b],x],x,a+b*ArcCos[c*x]] /;
FreeQ[{a,b,c,n},x]
```