Rules for normalizing integrands to known tangent forms

1. $\int u\,(c\,\text{Trig}[a+b\,x])^m\,(d\,\text{Trig}[a+b\,x])^n\,dx$ when KnownTangentIntegrandQ[u, x]

1: $\int u\,(c\,\text{Cot}[a+b\,x])^m\,(d\,\text{Tan}[a+b\,x])^n\,dx$ when KnownTangentIntegrandQ[u, x]

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x\left((c\,\text{Cot}[a+b\,x])^m\,(d\,\text{Tan}[a+b\,x])^m\right)=0$
- **Rule: If** KnownTangentIntegrandQ[u, x], **then**

$$\int u\,(c\,\text{Cot}[a+b\,x])^m\,(d\,\text{Tan}[a+b\,x])^n\,dx\;\longrightarrow\;(c\,\text{Cot}[a+b\,x])^m\,(d\,\text{Tan}[a+b\,x])^m\int u\,(d\,\text{Tan}[a+b\,x])^{n-m}\,dx$$

- **Program code:**

```
Int[u_*(c_.*cot[a_.+b_.*x_])^m_.*(d_.*tan[a_.+b_.*x_])^n_.,x_Symbol] :=
  (c*Cot[a+b*x])^m*(d*Tan[a+b*x])^m*Int[ActivateTrig[u]*(d*Tan[a+b*x])^(n-m),x] /;
FreeQ[{a,b,c,d,m,n},x] && KnownTangentIntegrandQ[u,x]
```

2: $\int u\,(c\,\text{Tan}[a+b\,x])^m\,(d\,\text{Cot}[a+b\,x])^n\,dx$ when KnownCotangentIntegrandQ[u, x]

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x\left((c\,\text{Tan}[a+b\,x])^m\,(d\,\text{Cot}[a+b\,x])^m\right)=0$
- **Rule: If** KnownCotangentIntegrandQ[u, x], **then**

$$\int u\,(c\,\text{Tan}[a+b\,x])^m\,(d\,\text{Cot}[a+b\,x])^n\,dx\;\longrightarrow\;(c\,\text{Tan}[a+b\,x])^m\,(d\,\text{Cot}[a+b\,x])^m\int u\,(d\,\text{Cot}[a+b\,x])^{n-m}\,dx$$

- **Program code:**

```
Int[u_*(c_.*tan[a_.+b_.*x_])^m_.*(d_.*cos[a_.+b_.*x_])^n_.,x_Symbol] :=
  (c*Tan[a+b*x])^m*(d*Cos[a+b*x])^m/(d*Sin[a+b*x])^m*Int[ActivateTrig[u]*(d*Sin[a+b*x])^m/(d*Cos[a+b*x])^(m-n),x] /;
FreeQ[{a,b,c,d,m,n},x] && KnownCotangentIntegrandQ[u,x]
```

2. $\int u\,(c\,\mathrm{Trig}[a+b\,x])^m\,dx$ when $m\notin\mathbb{Z}\,\land\,\mathrm{KnownTangentIntegrandQ}[u,x]$

1: $\int u\,(c\,\mathrm{Cot}[a+b\,x])^m\,dx$ when $m\notin\mathbb{Z}\,\land\,\mathrm{KnownTangentIntegrandQ}[u,x]$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((c\,\mathrm{Cot}[a+b\,x])^m\,(c\,\mathrm{Tan}[a+b\,x])^m\right)=0$

Rule: If $m\notin\mathbb{Z}\,\land\,\mathrm{KnownTangentIntegrandQ}[u,x]$**, then**

$$\int u\,(c\,\mathrm{Cot}[a+b\,x])^m\,dx\;\longrightarrow\;(c\,\mathrm{Cot}[a+b\,x])^m\,(c\,\mathrm{Tan}[a+b\,x])^m\int\frac{u}{(c\,\mathrm{Tan}[a+b\,x])^m}\,dx$$

Program code:

```
Int[u_*(c_.*cot[a_.+b_.*x_])^m_.,x_Symbol] :=
  (c*Cot[a+b*x])^m*(c*Tan[a+b*x])^m*Int[ActivateTrig[u]/(c*Tan[a+b*x])^m,x] /;
FreeQ[{a,b,c,m},x] && Not[IntegerQ[m]] && KnownTangentIntegrandQ[u,x]
```

2: $\int u\,(c\,\mathrm{Tan}[a+b\,x])^m\,dx$ when $m\notin\mathbb{Z}\,\land\,\mathrm{KnownCotangentIntegrandQ}[u,x]$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((c\,\mathrm{Cot}[a+b\,x])^m\,(c\,\mathrm{Tan}[a+b\,x])^m\right)=0$

Rule: If $m\notin\mathbb{Z}\,\land\,\mathrm{KnownCotangentIntegrandQ}[u,x]$**, then**

$$\int u\,(c\,\mathrm{Tan}[a+b\,x])^m\,dx\;\longrightarrow\;(c\,\mathrm{Cot}[a+b\,x])^m\,(c\,\mathrm{Tan}[a+b\,x])^m\int\frac{u}{(c\,\mathrm{Cot}[a+b\,x])^m}\,dx$$

Program code:

```
Int[u_*(c_.*tan[a_.+b_.*x_])^m_.,x_Symbol] :=
  (c*Cot[a+b*x])^m*(c*Tan[a+b*x])^m*Int[ActivateTrig[u]/(c*Cot[a+b*x])^m,x] /;
FreeQ[{a,b,c,m},x] && Not[IntegerQ[m]] && KnownCotangentIntegrandQ[u,x]
```

3. $\int u\,(A + B\,\text{Cot}[a + b\,x])\,dx$ when KnownTangentIntegrandQ[u, x]

1: $\int u\,(c\,\text{Tan}[a + b\,x])^n\,(A + B\,\text{Cot}[a + b\,x])\,dx$ when KnownTangentIntegrandQ[u, x]

Derivation: Algebraic normalization

Rule: If KnownTangentIntegrandQ[u, x], then

$$\int u\,(c\,\text{Tan}[a + b\,x])^n\,(A + B\,\text{Cot}[a + b\,x])\,dx \longrightarrow c\int u\,(c\,\text{Tan}[a + b\,x])^{n-1}\,(B + A\,\text{Tan}[a + b\,x])\,dx$$

Program code:

```
Int[u_*(c_.*tan[a_.+b_.*x_])^n_.*(A_+B_.*cot[a_.+b_.*x_]),x_Symbol] :=
  c*Int[ActivateTrig[u]*(c*Tan[a+b*x])^(n-1)*(B+A*Tan[a+b*x]),x] /;
FreeQ[{a,b,c,A,B,n},x] && KnownTangentIntegrandQ[u,x]
```

```
Int[u_*(c_.*cot[a_.+b_.*x_])^n_.*(A_+B_.*tan[a_.+b_.*x_]),x_Symbol] :=
  c*Int[ActivateTrig[u]*(c*Cot[a+b*x])^(n-1)*(B+A*Cot[a+b*x]),x] /;
FreeQ[{a,b,c,A,B,n},x] && KnownCotangentIntegrandQ[u,x]
```

2: $\int u\,(A + B\,\text{Cot}[a + b\,x])\,dx$ when KnownTangentIntegrandQ[u, x]

Derivation: Algebraic normalization

Rule: If KnownTangentIntegrandQ[u, x], then

$$\int u\,(A + B\,\text{Cot}[a + b\,x])\,dx \longrightarrow \int \frac{u\,(B + A\,\text{Tan}[a + b\,x])}{\text{Tan}[a + b\,x]}\,dx$$

Program code:

```
Int[u_*(A_+B_.*cot[a_.+b_.*x_]),x_Symbol] :=
  Int[ActivateTrig[u]*(B+A*Tan[a+b*x])/Tan[a+b*x],x] /;
FreeQ[{a,b,A,B},x] && KnownTangentIntegrandQ[u,x]
```

```
Int[u_*(A_+B_.*tan[a_.+b_.*x_]),x_Symbol] :=
  Int[ActivateTrig[u]*(B+A*Cot[a+b*x])/Cot[a+b*x],x] /;
FreeQ[{a,b,A,B},x] && KnownCotangentIntegrandQ[u,x]
```

4. $\int u\left(A+B\,\text{Cot}[a+b\,x]+C\,\text{Cot}[a+b\,x]^2\right)dx$ when **KnownTangentIntegrandQ[u, x]**

1: $\int u\,(c\,\text{Tan}[a+b\,x])^n\left(A+B\,\text{Cot}[a+b\,x]+C\,\text{Cot}[a+b\,x]^2\right)dx$ when **KnownTangentIntegrandQ[u, x]**

Derivation: Algebraic normalization

Rule: If KnownTangentIntegrandQ[u, x], then

$$\int u\,(c\,\text{Tan}[a+b\,x])^n\left(A+B\,\text{Cot}[a+b\,x]+C\,\text{Cot}[a+b\,x]^2\right)dx\;\longrightarrow\;c^2\int u\,(c\,\text{Tan}[a+b\,x])^{n-2}\left(C+B\,\text{Tan}[a+b\,x]+A\,\text{Tan}[a+b\,x]^2\right)dx$$

Program code:

```
Int[u_.*(c_.*tan[a_.+b_.*x_])^n_.*(A_.+B_.*cot[a_.+b_.*x_]+C_.*cot[a_.+b_.*x_]^2),x_Symbol] :=
  c^2*Int[ActivateTrig[u]*(c*Tan[a+b*x])^(n-2)*(C+B*Tan[a+b*x]+A*Tan[a+b*x]^2),x] /;
FreeQ[{a,b,c,A,B,C,n},x] && KnownTangentIntegrandQ[u,x]
```

```
Int[u_.*(c_.*cot[a_.+b_.*x_])^n_.*(A_.+B_.*tan[a_.+b_.*x_]+C_.*tan[a_.+b_.*x_]^2),x_Symbol] :=
  c^2*Int[ActivateTrig[u]*(c*Cot[a+b*x])^(n-2)*(C+B*Cot[a+b*x]+A*Cot[a+b*x]^2),x] /;
FreeQ[{a,b,c,A,B,C,n},x] && KnownCotangentIntegrandQ[u,x]
```

```
Int[u_.*(c_.*tan[a_.+b_.*x_])^n_.*(A_+C_.*cot[a_.+b_.*x_]^2),x_Symbol] :=
  c^2*Int[ActivateTrig[u]*(c*Tan[a+b*x])^(n-2)*(C+A*Tan[a+b*x]^2),x] /;
FreeQ[{a,b,c,A,C,n},x] && KnownTangentIntegrandQ[u,x]
```

```
Int[u_.*(c_.*cot[a_.+b_.*x_])^n_.*(A_+C_.*tan[a_.+b_.*x_]^2),x_Symbol] :=
  c^2*Int[ActivateTrig[u]*(c*Cot[a+b*x])^(n-2)*(C+A*Cot[a+b*x]^2),x] /;
FreeQ[{a,b,c,A,C,n},x] && KnownCotangentIntegrandQ[u,x]
```

2: $\int u\,(A + B\,\mathrm{Cot}[a + b\,x] + C\,\mathrm{Cot}[a + b\,x]^2)\,dx$ when **KnownTangentIntegrandQ[u, x]**

Derivation: Algebraic normalization

Rule: If KnownTangentIntegrandQ[u, x], then

$$\int u\,(A + B\,\mathrm{Cot}[a + b\,x] + C\,\mathrm{Cot}[a + b\,x]^2)\,dx \longrightarrow \int \frac{u\,(C + B\,\mathrm{Tan}[a + b\,x] + A\,\mathrm{Tan}[a + b\,x]^2)}{\mathrm{Tan}[a + b\,x]^2}\,dx$$

Program code:

```
Int[u_*(A_.+B_.*cot[a_.+b_.*x_]+C_.*cot[a_.+b_.*x_]^2),x_Symbol] :=
  Int[ActivateTrig[u]*(C+B*Tan[a+b*x]+A*Tan[a+b*x]^2)/Tan[a+b*x]^2,x] /;
FreeQ[{a,b,A,B,C},x] && KnownTangentIntegrandQ[u,x]
```

```
Int[u_*(A_.+B_.*tan[a_.+b_.*x_]+C_.*tan[a_.+b_.*x_]^2),x_Symbol] :=
  Int[ActivateTrig[u]*(C+B*Cot[a+b*x]+A*Cot[a+b*x]^2)/Cot[a+b*x]^2,x] /;
FreeQ[{a,b,A,B,C},x] && KnownCotangentIntegrandQ[u,x]
```

```
Int[u_*(A_+C_.*cot[a_.+b_.*x_]^2),x_Symbol] :=
  Int[ActivateTrig[u]*(C+A*Tan[a+b*x]^2)/Tan[a+b*x]^2,x] /;
FreeQ[{a,b,A,C},x] && KnownTangentIntegrandQ[u,x]
```

```
Int[u_*(A_+C_.*tan[a_.+b_.*x_]^2),x_Symbol] :=
  Int[ActivateTrig[u]*(C+A*Cot[a+b*x]^2)/Cot[a+b*x]^2,x] /;
FreeQ[{a,b,A,C},x] && KnownCotangentIntegrandQ[u,x]
```

5: $\int u\,(A + B\,\mathrm{Tan}[a + b\,x] + C\,\mathrm{Cot}[a + b\,x])\,dx$

- **Derivation: Algebraic normalization**
- **Rule:**

$$\int u\,(A + B\,\mathrm{Tan}[a + b\,x] + C\,\mathrm{Cot}[a + b\,x])\,dx \longrightarrow \int \frac{u\,\left(C + A\,\mathrm{Tan}[a + b\,x] + B\,\mathrm{Tan}[a + b\,x]^2\right)}{\mathrm{Tan}[a + b\,x]}\,dx$$

- **Program code:**

```
Int[u_*(A_.+B_.*tan[a_.+b_.*x_]+C_.*cot[a_.+b_.*x_]),x_Symbol] :=
  Int[ActivateTrig[u]*(C+A*Tan[a+b*x]+B*Tan[a+b*x]^2)/Tan[a+b*x],x] /;
FreeQ[{a,b,A,B,C},x]
```

6: $\int u\,\left(A\,\mathrm{Tan}[a + b\,x]^n + B\,\mathrm{Tan}[a + b\,x]^{n+1} + C\,\mathrm{Tan}[a + b\,x]^{n+2}\right)\,dx$

- **Derivation: Algebraic normalization**
- **Rule:**

$$\int u\,\left(A\,\mathrm{Tan}[a + b\,x]^n + B\,\mathrm{Tan}[a + b\,x]^{n+1} + C\,\mathrm{Tan}[a + b\,x]^{n+2}\right)\,dx \longrightarrow \int u\,\mathrm{Tan}[a + b\,x]^n\,\left(A + B\,\mathrm{Tan}[a + b\,x] + C\,\mathrm{Tan}[a + b\,x]^2\right)\,dx$$

- **Program code:**

```
Int[u_*(A_.*tan[a_.+b_.*x_]^n_.+B_.*tan[a_.+b_.*x_]^n1_+C_.*tan[a_.+b_.*x_]^n2_),x_Symbol] :=
  Int[ActivateTrig[u]*Tan[a+b*x]^n*(A+B*Tan[a+b*x]+C*Tan[a+b*x]^2),x] /;
FreeQ[{a,b,A,B,C,n},x] && EqQ[n1,n+1] && EqQ[n2,n+2]
```

```
Int[u_*(A_.*cot[a_.+b_.*x_]^n_.+B_.*cot[a_.+b_.*x_]^n1_+C_.*cot[a_.+b_.*x_]^n2_),x_Symbol] :=
  Int[ActivateTrig[u]*Cot[a+b*x]^n*(A+B*Cot[a+b*x]+C*Cot[a+b*x]^2),x] /;
FreeQ[{a,b,A,B,C,n},x] && EqQ[n1,n+1] && EqQ[n2,n+2]
```