Rules for integrands of the form $u\,(a + b\,\text{ArcSech}[c + d\,x])^p$

1. $\int (a + b\,\text{ArcSec}[c + d\,x])^p\,dx$

1: $\int \text{ArcSec}[c + d\,x]\,dx$

- **Reference: G&R 2.821.2, CRC 445, A&S 4.4.62**
- **Reference: G&R 2.821.1, CRC 446, A&S 4.4.61**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{ArcSec}[c + d\,x]\,dx \longrightarrow \frac{(c + d\,x)\,\text{ArcSec}[c + d\,x]}{d} - \int \frac{1}{(c + d\,x)\,\sqrt{1 - \frac{1}{(c+d\,x)^2}}}\,dx$$

- **Program code:**

```
Int[ArcSec[c_+d_.*x_],x_Symbol] :=
  (c+d*x)*ArcSec[c+d*x]/d -
  Int[1/((c+d*x)*Sqrt[1-1/(c+d*x)^2]),x] /;
FreeQ[{c,d},x]
```

```
Int[ArcCsc[c_+d_.*x_],x_Symbol] :=
  (c+d*x)*ArcCsc[c+d*x]/d +
  Int[1/((c+d*x)*Sqrt[1-1/(c+d*x)^2]),x] /;
FreeQ[{c,d},x]
```

2: $\int (a + b \,\mathrm{ArcSec}[c + d\,x])^p \,dx$ when $p \in \mathbb{Z}^+$

- **Derivation: Integration by substitution**
- **Rule: If $p \in \mathbb{Z}^+$, then**

$$\int (a + b \,\mathrm{ArcSec}[c + d\,x])^p \,dx \longrightarrow \frac{1}{d}\,\mathrm{Subst}\left[\int (a + b \,\mathrm{ArcSec}[x])^p \,dx,\ x,\ c + d\,x\right]$$

- **Program code:**

```
Int[(a_.+b_.*ArcSec[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[(a+b*ArcSec[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d},x] && IGtQ[p,0]
```

```
Int[(a_.+b_.*ArcCsc[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[(a+b*ArcCsc[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d},x] && IGtQ[p,0]
```

U: $\int (a + b \,\mathrm{ArcSec}[c + d\,x])^p \,dx$ when $p \notin \mathbb{Z}^+$

- **Rule: If $p \notin \mathbb{Z}^+$, then**

$$\int (a + b \,\mathrm{ArcSec}[c + d\,x])^p \,dx \longrightarrow \int (a + b \,\mathrm{ArcSec}[c + d\,x])^p \,dx$$

- **Program code:**

```
Int[(a_.+b_.*ArcSec[c_+d_.*x_])^p_,x_Symbol] :=
  Unintegrable[(a+b*ArcSec[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,p},x] && Not[IGtQ[p,0]]
```

```
Int[(a_.+b_.*ArcCsc[c_+d_.*x_])^p_,x_Symbol] :=
  Unintegrable[(a+b*ArcCsc[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,p},x] && Not[IGtQ[p,0]]
```

2. $\int (e+f x)^m (a+b\,\text{ArcSec}[c+d x])^p \, dx$

1: $\int (e+f x)^m (a+b\,\text{ArcSec}[c+d x])^p \, dx$ when $d e-c f == 0 \wedge p \in \mathbb{Z}^+$

Derivation: Integration by substitution

Rule: If $d e-c f == 0 \wedge p \in \mathbb{Z}^+$**, then**

$$\int (e+f x)^m (a+b\,\text{ArcSec}[c+d x])^p \, dx \longrightarrow \frac{1}{d}\,\text{Subst}\left[\int \left(\frac{f x}{d}\right)^m (a+b\,\text{ArcSec}[x])^p \, dx,\ x,\ c+d x\right]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcSec[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[(f*x/d)^m*(a+b*ArcSec[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[d*e-c*f,0] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcCsc[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[(f*x/d)^m*(a+b*ArcCsc[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[d*e-c*f,0] && IGtQ[p,0]
```

2: $\int (e+f x)^m (a+b\,\text{ArcSec}[c+d x])^p \, dx$ when $p \in \mathbb{Z}^+ \wedge m \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $m \in \mathbb{Z}$**, then**
$(e+f x)^m F[\text{ArcSec}[c+d x]] == \frac{1}{d^{m+1}}\,\text{Subst}[F[x]\,\text{Sec}[x]\,\text{Tan}[x]\,(d e-c f+f\,\text{Sec}[x])^m,\ x,\ \text{ArcSec}[c+d x]]\ \partial_x \text{ArcSec}[c+d x]$

Rule: If $p \in \mathbb{Z}^+ \wedge m \in \mathbb{Z}$**, then**

$$\int (e+f x)^m (a+b\,\text{ArcSec}[c+d x])^p \, dx \longrightarrow \frac{1}{d^{m+1}}\,\text{Subst}\left[\int (a+b x)^p\,\text{Sec}[x]\,\text{Tan}[x]\,(d e-c f+f\,\text{Sec}[x])^m \, dx,\ x,\ \text{ArcSec}[c+d x]\right]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcSec[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d^(m+1)*Subst[Int[(a+b*x)^p*Sec[x]*Tan[x]*(d*e-c*f+f*Sec[x])^m,x],x,ArcSec[c+d*x]] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && IntegerQ[m]
```

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcCsc[c_+d_.*x_])^p_.,x_Symbol] :=
  -1/d^(m+1)*Subst[Int[(a+b*x)^p*Csc[x]*Cot[x]*(d*e-c*f+f*Csc[x])^m,x],x,ArcCsc[c+d*x]] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && IntegerQ[m]
```

3: $\int (e+f\,x)^m\,(a+b\,\text{ArcSec}[c+d\,x])^p\,dx$ when $p\in\mathbb{Z}^+$

Derivation: Integration by substitution

Rule: If $p\in\mathbb{Z}^+$, then

$$\int (e+f\,x)^m\,(a+b\,\text{ArcSec}[c+d\,x])^p\,dx \longrightarrow \frac{1}{d}\,\text{Subst}\Big[\int\left(\frac{d\,e-c\,f}{d}+\frac{f\,x}{d}\right)^m(a+b\,\text{ArcSec}[x])^p\,dx,\ x,\ c+d\,x\Big]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcSec[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(a+b*ArcSec[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcCsc[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(a+b*ArcCsc[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[p,0]
```

U: $\int (e+f\,x)^m\,(a+b\,\text{ArcSec}[c+d\,x])^p\,dx$ when $p\notin\mathbb{Z}^+$

Rule: If $p\notin\mathbb{Z}^+$, then

$$\int (e+f\,x)^m\,(a+b\,\text{ArcSec}[c+d\,x])^p\,dx \longrightarrow \int (e+f\,x)^m\,(a+b\,\text{ArcSec}[c+d\,x])^p\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcSec[c_+d_.*x_])^p_,x_Symbol] :=
  Unintegrable[(e+f*x)^m*(a+b*ArcSec[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && Not[IGtQ[p,0]]
```

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcCsc[c_+d_.*x_])^p_,x_Symbol] :=
  Unintegrable[(e+f*x)^m*(a+b*ArcCsc[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && Not[IGtQ[p,0]]
```

Rules for integrands involving inverse secants and cosecants

1: $\int u\,\mathrm{ArcSec}\left[\frac{c}{a+b\,x^n}\right]^m dx$

- **Derivation: Algebraic simplification**
- **Basis:** $\mathrm{ArcSec}[z] == \mathrm{ArcCos}\left[\frac{1}{z}\right]$
- **Rule:**

$$\int u\,\mathrm{ArcSec}\left[\frac{c}{a+b\,x^n}\right]^m dx \longrightarrow \int u\,\mathrm{ArcCos}\left[\frac{a}{c}+\frac{b\,x^n}{c}\right]^m dx$$

- **Program code:**

```
Int[u_.*ArcSec[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcCos[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

```
Int[u_.*ArcCsc[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcSin[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

2: $\int u\,f^{c\,\mathrm{ArcSec}[a+b\,x]^n}\,dx$

- **Derivation: Integration by substitution**
- **Basis:** $F[x,\ \mathrm{ArcSec}[a+b\,x]] == \frac{1}{b}\,\mathrm{Subst}\left[F\left[-\frac{a}{b}+\frac{\mathrm{Sec}[x]}{b},\ x\right]\mathrm{Sec}[x]\,\mathrm{Tan}[x],\ x,\ \mathrm{ArcSec}[a+b\,x]\right]\partial_x\mathrm{ArcSec}[a+b\,x]$
- **Rule: If** $n\in\mathbb{Z}^+$, **then**

$$\int u\,f^{c\,\mathrm{ArcSec}[a+b\,x]^n}\,dx \longrightarrow \frac{1}{b}\,\mathrm{Subst}\left[\int \mathrm{Subst}\left[u,\ x,\ -\frac{a}{b}+\frac{\mathrm{Sec}[x]}{b}\right]f^{c\,x^n}\,\mathrm{Sec}[x]\,\mathrm{Tan}[x]\,dx,\ x,\ \mathrm{ArcSec}[a+b\,x]\right]$$

- **Program code:**

```
Int[u_.*f_^(c_.*ArcSec[a_.+b_.*x_]^n_.),x_Symbol] :=
  1/b*Subst[Int[ReplaceAll[u,x->-a/b+Sec[x]/b]*f^(c*x^n)*Sec[x]*Tan[x],x],x,ArcSec[a+b*x]] /;
FreeQ[{a,b,c,f},x] && IGtQ[n,0]
```

```
Int[u_.*f_^(c_.*ArcCsc[a_.+b_.*x_]^n_.),x_Symbol] :=
  -1/b*Subst[Int[ReplaceAll[u,x→-a/b+Csc[x]/b]*f^(c*x^n)*Csc[x]*Cot[x],x],x,ArcCsc[a+b*x]] /;
FreeQ[{a,b,c,f},x] && IGtQ[n,0]
```

3. $\int v\,(a + b\,\text{ArcSec}[u])\,dx$ when u is free of inverse functions

1: $\int \text{ArcSec}[u]\,dx$ when u is free of inverse functions

Derivation: Integration by parts and piecewise constant extraction

- **Basis:** $\partial_x \text{ArcSec}[F[x]] == \frac{\partial_x F[x]}{\sqrt{F[x]^2}\,\sqrt{F[x]^2-1}}$

- **Basis:** $\partial_x \frac{F[x]}{\sqrt{F[x]^2}} == 0$

Rule: If u is free of inverse functions, then

$$\int \text{ArcSec}[u]\,dx \longrightarrow x\,\text{ArcSec}[u] - \int \frac{x\,\partial_x u}{\sqrt{u^2}\,\sqrt{u^2-1}}\,dx \longrightarrow x\,\text{ArcSec}[u] - \frac{u}{\sqrt{u^2}} \int \frac{x\,\partial_x u}{u\,\sqrt{u^2-1}}\,dx$$

Program code:

```
Int[ArcSec[u_],x_Symbol] :=
  x*ArcSec[u] -
  u/Sqrt[u^2]*Int[SimplifyIntegrand[x*D[u,x]/(u*Sqrt[u^2-1]),x],x] /;
InverseFunctionFreeQ[u,x] && Not[FunctionOfExponentialQ[u,x]]
```

```
Int[ArcCsc[u_],x_Symbol] :=
  x*ArcCsc[u] +
  u/Sqrt[u^2]*Int[SimplifyIntegrand[x*D[u,x]/(u*Sqrt[u^2-1]),x],x] /;
InverseFunctionFreeQ[u,x] && Not[FunctionOfExponentialQ[u,x]]
```

2: $\int (c+d\,x)^m\,(a+b\,\text{ArcSec}[u])\,dx$ when $m \neq -1 \land u$ is free of inverse functions

Derivation: Integration by parts and piecewise constant extraction

- **Basis:** $\partial_x\,(a+b\,\text{ArcSec}[F[x]]) = \frac{b\,\partial_x F[x]}{\sqrt{F[x]^2}\,\sqrt{F[x]^2-1}}$
- **Basis:** $\partial_x\,\frac{F[x]}{\sqrt{F[x]^2}} = 0$

Rule: If $m \neq -1 \land u$ is free of inverse functions, then

$$\int (c+d\,x)^m\,(a+b\,\text{ArcSec}[u])\,dx \to \frac{(c+d\,x)^{m+1}\,(a+b\,\text{ArcSec}[u])}{d\,(m+1)} - \frac{b}{d\,(m+1)}\int \frac{(c+d\,x)^{m+1}\,\partial_x u}{\sqrt{u^2}\,\sqrt{u^2-1}}\,dx$$

$$\to \frac{(c+d\,x)^{m+1}\,(a+b\,\text{ArcSec}[u])}{d\,(m+1)} - \frac{b\,u}{d\,(m+1)\,\sqrt{u^2}}\int \frac{(c+d\,x)^{m+1}\,\partial_x u}{u\,\sqrt{u^2-1}}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_.+b_.*ArcSec[u_]),x_Symbol] :=
  (c+d*x)^(m+1)*(a+b*ArcSec[u])/(d*(m+1)) -
  b*u/(d*(m+1)*Sqrt[u^2])*Int[SimplifyIntegrand[(c+d*x)^(m+1)*D[u,x]/(u*Sqrt[u^2-1]),x],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1] && InverseFunctionFreeQ[u,x] && Not[FunctionOfQ[(c+d*x)^(m+1),u,x]] && Not[FunctionOfExponential(
```

```
Int[(c_.+d_.*x_)^m_.*(a_.+b_.*ArcCsc[u_]),x_Symbol] :=
  (c+d*x)^(m+1)*(a+b*ArcCsc[u])/(d*(m+1)) +
  b*u/(d*(m+1)*Sqrt[u^2])*Int[SimplifyIntegrand[(c+d*x)^(m+1)*D[u,x]/(u*Sqrt[u^2-1]),x],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1] && InverseFunctionFreeQ[u,x] && Not[FunctionOfQ[(c+d*x)^(m+1),u,x]] && Not[FunctionOfExponential(
```

3: $\int v\,(a+b\,\text{ArcSec}[u])\,dx$ when u and $\int v\,dx$ are free of inverse functions

- **Derivation: Integration by parts and piecewise constant extraction**
- **Basis:** $\partial_x\,(a+b\,\text{ArcSec}[F[x]]) = \frac{b\,\partial_x F[x]}{\sqrt{F[x]^2}\,\sqrt{F[x]^2-1}}$
- **Basis:** $\partial_x\,\frac{F[x]}{\sqrt{F[x]^2}} = 0$
- **Rule: If u is free of inverse functions, let $w \to \int v\,dx$, if w is free of inverse functions, then**

$$\int v\,(a+b\,\text{ArcSec}[u])\,dx \longrightarrow w\,(a+b\,\text{ArcSec}[u]) - b\int \frac{w\,\partial_x u}{\sqrt{u^2}\,\sqrt{u^2-1}}\,dx \longrightarrow w\,(a+b\,\text{ArcSec}[u]) - \frac{b\,u}{\sqrt{u^2}}\int \frac{w\,\partial_x u}{u\,\sqrt{u^2-1}}\,dx$$

- **Program code:**

```
Int[v_*(a_.+b_.*ArcSec[u_]),x_Symbol] :=
  With[{w=IntHide[v,x]},
  Dist[(a+b*ArcSec[u]),w,x] - b*u/Sqrt[u^2]*Int[SimplifyIntegrand[w*D[u,x]/(u*Sqrt[u^2-1]),x],x] /;
 InverseFunctionFreeQ[w,x]] /;
FreeQ[{a,b},x] && InverseFunctionFreeQ[u,x] && Not[MatchQ[v, (c_.+d_.*x)^m_. /; FreeQ[{c,d,m},x]]]
```

```
Int[v_*(a_.+b_.*ArcCsc[u_]),x_Symbol] :=
  With[{w=IntHide[v,x]},
  Dist[(a+b*ArcCsc[u]),w,x] + b*u/Sqrt[u^2]*Int[SimplifyIntegrand[w*D[u,x]/(u*Sqrt[u^2-1]),x],x] /;
 InverseFunctionFreeQ[w,x]] /;
FreeQ[{a,b},x] && InverseFunctionFreeQ[u,x] && Not[MatchQ[v, (c_.+d_.*x)^m_. /; FreeQ[{c,d,m},x]]]
```