Rules for integrands of the form $P_q[x]\left(a+b\,x+c\,x^2\right)^p$ when $q>1$

1: $\int P_q[x]\left(a+b\,x+c\,x^2\right)^p dx$ when $p+2\in\mathbb{Z}^+$

- **Derivation: Algebraic expansion**
- **Rule 1.2.1.8.1: If $p+2\in\mathbb{Z}^+$, then**

$$\int P_q[x]\left(a+b\,x+c\,x^2\right)^p dx \longrightarrow \int \text{ExpandIntegrand}\left[P_q[x]\left(a+b\,x+c\,x^2\right)^p,\,x\right] dx$$

- **Program code:**

```
Int[Pq_*(a_+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[Pq*(a+b*x+c*x^2)^p,x],x] /;
FreeQ[{a,b,c},x] && PolyQ[Pq,x] && IGtQ[p,-2]
```

2: $\int P_q[x]\left(a+b\,x+c\,x^2\right)^p dx$ when $P_q[x,0]==0$

- **Derivation: Algebraic simplification**
- **Rule 1.2.1.8.2: If $P_q[x,0]==0$, then**

$$\int P_q[x]\left(a+b\,x+c\,x^2\right)^p dx \longrightarrow \int x\,\text{PolynomialQuotient}\left[P_q[x],\,x,\,x\right]\left(a+b\,x+c\,x^2\right)^p dx$$

- **Program code:**

```
Int[Pq_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[x*PolynomialQuotient[Pq,x,x]*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,p},x] && PolyQ[Pq,x] && EqQ[Coeff[Pq,x,0],0] && Not[MatchQ[Pq,x^m_.*u_. /; IntegerQ[m]]]
```

3: $\int P_q[x] \left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c==0$

- **Derivation: Piecewise constant extraction**
- **Basis: If $b^2-4\,a\,c==0$, then $\partial_x \frac{\left(a+b\,x+c\,x^2\right)^p}{(b+2\,c\,x)^{2\,p}}==0$**
- **Rule 1.2.1.8.3: If $b^2-4\,a\,c==0$, then**

$$\int P_q[x] \left(a+b\,x+c\,x^2\right)^p\,dx \longrightarrow \frac{\left(a+b\,x+c\,x^2\right)^{\text{FracPart}[p]}}{(4\,c)^{\text{IntPart}[p]}\,(b+2\,c\,x)^{2\,\text{FracPart}[p]}} \int P_q[x] \left(b+2\,c\,x^2\right)^{2\,p}\,dx$$

- **Program code:**

```
Int[Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (a+b*x+c*x^2)^FracPart[p]/((4*c)^IntPart[p]*(b+2*c*x)^(2*FracPart[p]))*Int[Pq*(b+2*c*x)^(2*p),x] /;
FreeQ[{a,b,c,p},x] && PolyQ[Pq,x] && EqQ[b^2-4*a*c,0]
```

4: $\int P_q[x]\,(a+b\,x+c\,x^2)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,p<-1$

Derivation: Algebraic expansion and quadratic recurrence 2a

- **Rule 1.2.1.8.4: If $b^2-4\,a\,c\neq 0\,\wedge\,p<-1$,**
  let $Q_{q-2}[x]\to \text{PolynomialQuotient}\left[P_q[x],\ a+b\,x+c\,x^2,\ x\right]$ and $f+g\,x\to \text{PolynomialRemainder}\left[P_q[x],\ a+b\,x+c\,x^2,\ x\right]$, then

$$\int P_q[x]\,(a+b\,x+c\,x^2)^p\,dx\ \longrightarrow$$

$$\int (f+g\,x)\,(a+b\,x+c\,x^2)^p\,dx+\int Q_{q-2}[x]\,(a+b\,x+c\,x^2)^{p+1}\,dx\ \longrightarrow$$

$$\frac{(b\,f-2\,a\,g+(2\,c\,f-b\,g)\,x)\,(a+b\,x+c\,x^2)^{p+1}}{(p+1)\,(b^2-4\,a\,c)}+\frac{1}{(p+1)\,(b^2-4\,a\,c)}\int (a+b\,x+c\,x^2)^{p+1}\left((p+1)\,(b^2-4\,a\,c)\,Q_{q-2}[x]-(2\,p+3)\,(2\,c\,f-b\,g)\right)dx$$

- **Program code:**

```
Int[Pq_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a+b*x+c*x^2,x],
        f=Coeff[PolynomialRemainder[Pq,a+b*x+c*x^2,x],x,0],
        g=Coeff[PolynomialRemainder[Pq,a+b*x+c*x^2,x],x,1]},
  (b*f-2*a*g+(2*c*f-b*g)*x)*(a+b*x+c*x^2)^(p+1)/((p+1)*(b^2-4*a*c)) +
  1/((p+1)*(b^2-4*a*c))*Int[(a+b*x+c*x^2)^(p+1)*ExpandToSum[(p+1)*(b^2-4*a*c)*Q-(2*p+3)*(2*c*f-b*g),x],x]] /;
FreeQ[{a,b,c},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1]
```

5: $\int P_q[x]\,(a+b\,x+c\,x^2)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,p\nless -1$

- **Reference: G&R 2.160.3**
- **Derivation: Trinomial recurrence 3a with $A=0$, $B=1$ and $m=m-n$**
- **Reference: G&R 2.104**
- **Note: This special case of the Ostrogradskiy-Hermite integration method reduces the degree of the polynomial in the resulting integrand.**
- **Rule 1.2.1.8.5: If $b^2-4\,a\,c\neq 0\,\wedge\,p\nless -1$, let $e\to P_q[x,\ q]$, then**

$$\int P_q[x]\,(a+b\,x+c\,x^2)^p\,dx\ \longrightarrow$$

$$\int \left(P_q[x] - e\,x^q\right) \left(a + b\,x + c\,x^2\right)^p \,dx + e \int x^q \left(a + b\,x + c\,x^2\right)^p \,dx \longrightarrow$$

$$\frac{e\,x^{q-1} \left(a + b\,x + c\,x^2\right)^{p+1}}{c\,(q + 2\,p + 1)} + \frac{1}{c\,(q + 2\,p + 1)} \int \left(a + b\,x + c\,x^2\right)^p \left(c\,(q + 2\,p + 1)\,P_q[x] - a\,e\,(q - 1)\,x^{q-2} - b\,e\,(q + p)\,x^{q-1} - c\,e\,(q + 2\,p + 1)\,x^q\right) dx$$

- **Program code:**

```
Int[Pq_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x],e=Coeff[Pq,x,Expon[Pq,x]]},
  e*x^(q-1)*(a+b*x+c*x^2)^(p+1)/(c*(q+2*p+1)) +
  1/(c*(q+2*p+1))*Int[(a+b*x+c*x^2)^p*
     ExpandToSum[c*(q+2*p+1)*Pq-a*e*(q-1)*x^(q-2)-b*e*(q+p)*x^(q-1)-c*e*(q+2*p+1)*x^q,x],x]] /;
FreeQ[{a,b,c,p},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && Not[LeQ[p,-1]]
```