Rules for integrands of the form $(f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p$

0: $\int x^m \left(d+\frac{e}{x}\right)^q (a+b\,\text{Log}[c\,x^n])^p\,dx$ when $m==q\land q\in\mathbb{Z}$

- **Derivation: Algebraic simplification**
- **Rule: If $m==q\land q\in\mathbb{Z}$, then**

$$\int x^m \left(d+\frac{e}{x}\right)^q (a+b\,\text{Log}[c\,x^n])^p\,dx\;\longrightarrow\;\int (e+d\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$$

- **Program code:**

```
Int[x_^m_.*(d_+e_./x_)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Int[(e+d*x)^q*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[m,q] && IntegerQ[q]
```

1: $\int x^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])\,dx$ when $q\in\mathbb{Z}^+\land m\in\mathbb{Z}$

- **Derivation: Integration by parts**
- **Basis:** $\partial_x\,(a+b\,\text{Log}[c\,x^n])==\frac{b\,n}{x}$
- **Rule: If $q\in\mathbb{Z}^+\land m\in\mathbb{Z}$, let $u\to\int x^m\,(d+e\,x^r)^q\,dx$, then**

$$\int x^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])\,dx\;\longrightarrow\;u\,(a+b\,\text{Log}[c\,x^n])-b\,n\int\frac{u}{x}\,dx$$

- **Program code:**

```
Int[x_^m_.*(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  With[{u=IntHide[x^m*(d+e*x^r)^q,x]},
  u*(a+b*Log[c*x^n]) - b*n*Int[SimplifyIntegrand[u/x,x],x]] /;
FreeQ[{a,b,c,d,e,n,r},x] && IGtQ[q,0] && IGtQ[m,0]
```

```
Int[x_^m_.*(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  With[{u=IntHide[x^m*(d+e*x^r)^q,x]},
  Dist[(a+b*Log[c*x^n]),u] - b*n*Int[SimplifyIntegrand[u/x,x],x]] /;
FreeQ[{a,b,c,d,e,n,r},x] && IGtQ[q,0] && IntegerQ[m] && Not[EqQ[q,1] && EqQ[m,-1]]
```

2: $\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\mathrm{Log}[c\,x^n])\,dx$ when $m+r\,(q+1)+1=0\,\wedge\,m\neq -1$

Derivation: Integration by parts

- **Basis: If** $m+r\,(q+1)+1=0\,\wedge\,m\neq -1$, **then** $(f\,x)^m\,(d+e\,x^r)^q = \partial_x \frac{(f\,x)^{m+1}\,(d+e\,x^r)^{q+1}}{d\,f\,(m+1)}$

Rule: If $m+r\,(q+1)+1=0\,\wedge\,m\neq -1$, **then**

$$\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\mathrm{Log}[c\,x^n])\,dx \longrightarrow \frac{(f\,x)^{m+1}\,(d+e\,x^r)^{q+1}\,(a+b\,\mathrm{Log}[c\,x^n])}{d\,f\,(m+1)} - \frac{b\,n}{d\,(m+1)}\int (f\,x)^m\,(d+e\,x^r)^{q+1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^r_.)^q_*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^r)^(q+1)*(a+b*Log[c*x^n])/(d*f*(m+1)) -
  b*n/(d*(m+1))*Int[(f*x)^m*(d+e*x^r)^(q+1),x] /;
FreeQ[{a,b,c,d,e,f,m,n,q,r},x] && EqQ[m+r*(q+1)+1,0] && NeQ[m,-1]
```

3. $\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\mathrm{Log}[c\,x^n])^p\,dx$ when $m=r-1\,\wedge\,p\in\mathbb{Z}^+$

1. $\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\mathrm{Log}[c\,x^n])^p\,dx$ when $m=r-1\,\wedge\,p\in\mathbb{Z}^+\,\wedge\,(m\in\mathbb{Z}\,\vee\,f>0)$

1: $\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\mathrm{Log}[c\,x^n])^p\,dx$ when $m=r-1\,\wedge\,p\in\mathbb{Z}^+\,\wedge\,(m\in\mathbb{Z}\,\vee\,f>0)\,\wedge\,r=n$

Derivation: Integration by substitution

Rule: If $m=r-1\,\wedge\,p\in\mathbb{Z}^+\,\wedge\,(m\in\mathbb{Z}\,\vee\,f>0)\,\wedge\,r=n$, **then**

$$\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\mathrm{Log}[c\,x^n])^p\,dx \longrightarrow \frac{f^m}{n}\,\mathrm{Subst}\Big[\int (d+e\,x)^q\,(a+b\,\mathrm{Log}[c\,x])^p\,dx,\ x,\ x^n\Big]$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^r_)^q_.*(a_.+b_.*Log[c_.*x_^n_])^p_.,x_Symbol] :=
  f^m/n*Subst[Int[(d+e*x)^q*(a+b*Log[c*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,e,f,m,n,q,r},x] && EqQ[m,r-1] && IGtQ[p,0] && (IntegerQ[m] || GtQ[f,0]) && EqQ[r,n]
```

2. $\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\mathrm{Log}[c\,x^n])^p\,dx$ when $m=r-1\,\wedge\,p\in\mathbb{Z}^+\,\wedge\,(m\in\mathbb{Z}\,\vee\,f>0)\,\wedge\,r\neq n$

1: $\int \frac{(f\,x)^m\,(a+b\,\text{Log}[c\,x^n])^p}{d+e\,x^r}\,dx$ when $m == r-1\,\wedge\,p\in\mathbb{Z}^+\wedge\,(m\in\mathbb{Z}\,\vee\,f>0)\,\wedge\,r\neq n$

Derivation: Integration by parts

Basis: $\frac{(f\,x)^m}{d+e\,x^r} == \frac{f^m}{e\,r}\,\partial_x\,\text{Log}\left[1+\frac{e\,x^r}{d}\right]$

Rule: If $m == r-1\,\wedge\,p\in\mathbb{Z}^+\wedge\,(m\in\mathbb{Z}\,\vee\,f>0)\,\wedge\,r\neq n$**, then**

$$\int \frac{(f\,x)^m\,(a+b\,\text{Log}[c\,x^n])^p}{d+e\,x^r}\,dx \longrightarrow \frac{f^m\,\text{Log}\left[1+\frac{e\,x^r}{d}\right]\,(a+b\,\text{Log}[c\,x^n])^p}{e\,r}-\frac{b\,f^m\,n\,p}{e\,r}\int \frac{\text{Log}\left[1+\frac{e\,x^r}{d}\right]\,(a+b\,\text{Log}[c\,x^n])^{p-1}}{x}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(a_.+b_.*Log[c_.*x_^n_.])^p_./(d_+e_.*x_^r_),x_Symbol] :=
  f^m*Log[1+e*x^r/d]*(a+b*Log[c*x^n])^p/(e*r) -
  b*f^m*n*p/(e*r)*Int[Log[1+e*x^r/d]*(a+b*Log[c*x^n])^(p-1)/x,x] /;
FreeQ[{a,b,c,d,e,f,m,n,r},x] && EqQ[m,r-1] && IGtQ[p,0] && (IntegerQ[m] || GtQ[f,0]) && NeQ[r,n]
```

2: $\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$ when $m == r-1\,\wedge\,p\in\mathbb{Z}^+\wedge\,(m\in\mathbb{Z}\,\vee\,f>0)\,\wedge\,r\neq n\,\wedge\,q\neq -1$

Derivation: Integration by parts

Rule: If $m == r-1\,\wedge\,p\in\mathbb{Z}^+\wedge\,(m\in\mathbb{Z}\,\vee\,f>0)\,\wedge\,r\neq n\,\wedge\,q\neq -1$**, then**

$$\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx \longrightarrow \frac{f^m\,(d+e\,x^r)^{q+1}\,(a+b\,\text{Log}[c\,x^n])^p}{e\,r\,(q+1)}-\frac{b\,f^m\,n\,p}{e\,r\,(q+1)}\int \frac{(d+e\,x^r)^{q+1}\,(a+b\,\text{Log}[c\,x^n])^{p-1}}{x}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^r_)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  f^m*(d+e*x^r)^(q+1)*(a+b*Log[c*x^n])^p/(e*r*(q+1)) -
  b*f^m*n*p/(e*r*(q+1))*Int[(d+e*x^r)^(q+1)*(a+b*Log[c*x^n])^(p-1)/x,x] /;
FreeQ[{a,b,c,d,e,f,m,n,q,r},x] && EqQ[m,r-1] && IGtQ[p,0] && (IntegerQ[m] || GtQ[f,0]) && NeQ[r,n] && NeQ[q,-1]
```

2: $\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$ when $m == r-1\,\wedge\,p\in\mathbb{Z}^+\wedge\,\neg\,(m\in\mathbb{Z}\,\vee\,f>0)$

Derivation: Piecewise constant extraction

Rule: If $m == r-1\,\wedge\,p\in\mathbb{Z}^+\wedge\,\neg\,(m\in\mathbb{Z}\,\vee\,f>0)$**, then**

$$\int (f x)^m (d + e x^r)^q (a + b \operatorname{Log}[c x^n])^p \, dx \longrightarrow \frac{(f x)^m}{x^m} \int x^m (d + e x^r)^q (a + b \operatorname{Log}[c x^n])^p \, dx$$

Program code:

```
Int[(f_*x_)^m_.*(d_+e_.*x_^r_)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  (f*x)^m/x^m*Int[x^m*(d+e*x^r)^q*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,q,r},x] && EqQ[m,r-1] && IGtQ[p,0] && Not[(IntegerQ[m] || GtQ[f,0])]
```

?. $\int \frac{x^m (a + b \operatorname{Log}[c x^n])^p}{d + e x^r} \, dx$ when $p \in \mathbb{Z}^+ \land r \in \mathbb{Z}^+ \land m \in \mathbb{Z}$

x: $\int \frac{x^m (a + b \operatorname{Log}[c x^n])^p}{d + e x^r} \, dx$ when $p \in \mathbb{Z}^+ \land r \in \mathbb{Z}^+ \land m - r + 1 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

- **Basis:** $\frac{x^m}{d+e x^r} == \frac{x^{m-r}}{e} - \frac{d x^{m-r}}{e (d+e x^r)}$

- **Rule: If** $p \in \mathbb{Z}^+ \land r \in \mathbb{Z}^+ \land m - r + 1 \in \mathbb{Z}^+$**, then**

$$\int \frac{x^m (a + b \operatorname{Log}[c x^n])^p}{d + e x^r} \, dx \longrightarrow \frac{1}{e} \int x^{m-r} (a + b \operatorname{Log}[c x^n])^p \, dx - \frac{d}{e} \int \frac{x^{m-r} (a + b \operatorname{Log}[c x^n])^p}{d + e x^r} \, dx$$

Program code:

```
(* Int[x_^m_.*(a_.+b_.*Log[c_.*x_^n_.])^p_./(d_+e_.*x_^r_.),x_Symbol] :=
  1/e*Int[x^(m-r)*(a+b*Log[c*x^n])^p,x] -
  d/e*Int[(x^(m-r)*(a+b*Log[c*x^n])^p)/(d+e*x^r),x] /;
FreeQ[{a,b,c,d,e,m,n,r},x] && IGtQ[p,0] && IGtQ[r,0] && IGeQ[m-r,0] *)
```

2. $\int \frac{x^m (a + b \text{Log}[c x^n])^p}{d + e x^r} \, dx$ when $p \in \mathbb{Z}^+ \land r \in \mathbb{Z}^+ \land m \in \mathbb{Z}^-$

1. $\int \frac{(a + b \text{Log}[c x^n])^p}{x (d + e x^r)} \, dx$ when $p \in \mathbb{Z}^+ \land r \in \mathbb{Z}^+$

1: $\int \frac{a + b \text{Log}[c x^n]}{x (d + e x^r)} \, dx$ when $\frac{r}{n} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $\frac{F[x^n]}{x} == \frac{1}{n} \text{Subst}\left[\frac{F[x]}{x}, x, x^n\right] \partial_x x^n$

Rule: If $\frac{r}{n} \in \mathbb{Z}$, then

$$\int \frac{a + b \text{Log}[c x^n]}{x (d + e x^r)} \, dx \longrightarrow \frac{1}{n} \text{Subst}\left[\int \frac{a + b \text{Log}[c x]}{x \left(d + e x^{r/n}\right)} \, dx, x, x^n\right]$$

Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_])/(x_*(d_+e_.*x_^r_.)),x_Symbol] :=
  1/n*Subst[Int[(a+b*Log[c*x])/(x*(d+e*x^(r/n))),x],x,x^n] /;
FreeQ[{a,b,c,d,e,n,r},x] && IntegerQ[r/n]
```

x: $\int \frac{(a + b \text{Log}[c x^n])^p}{x (d + e x)} \, dx$ when $p \in \mathbb{Z}^+$

Rule: Algebraic expansion

Basis: $\frac{1}{x (d+e x)} == \frac{1}{d x} - \frac{e}{d (d+e x)}$

Note: This rule returns antiderivative in terms of $\frac{e x}{d}$ instead of $\frac{d}{e x}$, but requires more steps and one more term.

Rule: If $p \in \mathbb{Z}^+$, then

$$\int \frac{(a + b \text{Log}[c x^n])^p}{x (d + e x)} \, dx \longrightarrow \frac{1}{d} \int \frac{(a + b \text{Log}[c x^n])^p}{x} \, dx - \frac{e}{d} \int \frac{(a + b \text{Log}[c x^n])^p}{d + e x} \, dx$$

Program code:

```
(* Int[(a_.+b_.*Log[c_.*x_^n_.])^p_./(x_*(d_+e_.*x_)),x_Symbol] :=
  1/d*Int[(a+b*Log[c*x^n])^p/x,x] - e/d*Int[(a+b*Log[c*x^n])^p/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[p,0] *)
```

x: $\int \frac{(a + b \operatorname{Log}[c\, x^n])^p}{x\, (d + e\, x^r)}\, dx$ when $p \in \mathbb{Z}^+$

Rule: Integration by parts

- **Basis:** $\frac{1}{x\,(d+e\,x^r)} == \partial_x \frac{r \operatorname{Log}[x] - \operatorname{Log}\left[1+\frac{e\,x^r}{d}\right]}{d\,r}$
- **Basis:** $\partial_x\, (a + b \operatorname{Log}[c\, x^n])^p == \frac{b\,n\,p\,(a+b \operatorname{Log}[c\,x^n])^{p-1}}{x}$

Note: This rule returns antiderivatives in terms of x^r instead of x^{-r}, but requires more steps and larger antiderivatives.

Rule: If $p \in \mathbb{Z}^+$, then

$$\int \frac{(a + b \operatorname{Log}[c\, x^n])^p}{x\, (d + e\, x^r)}\, dx \longrightarrow$$

$$\frac{\left(r \operatorname{Log}[x] - \operatorname{Log}\left[1 + \frac{e\,x^r}{d}\right]\right) (a + b \operatorname{Log}[c\, x^n])^p}{d\, r} - \frac{b\,n\,p}{d} \int \frac{\operatorname{Log}[x]\, (a + b \operatorname{Log}[c\, x^n])^{p-1}}{x}\, dx + \frac{b\,n\,p}{d\,r} \int \frac{\operatorname{Log}\left[1 + \frac{e\,x^r}{d}\right] (a + b \operatorname{Log}[c\, x^n])^{p-1}}{x}\, dx$$

Program code:

```
(* Int[(a_.+b_.*Log[c_.*x_^n_.])^p_./(x_*(d_+e_.*x_^r_.)),x_Symbol] :=
  (r*Log[x]-Log[1+(e*x^r)/d])*(a+b*Log[c*x^n])^p/(d*r) -
  b*n*p/d*Int[Log[x]*(a+b*Log[c*x^n])^(p-1)/x,x] +
  b*n*p/(d*r)*Int[Log[1+(e*x^r)/d]*(a+b*Log[c*x^n])^(p-1)/x,x] /;
FreeQ[{a,b,c,d,e,n,r},x] && IGtQ[p,0] *)
```

2: $\int \frac{(a + b \operatorname{Log}[c x^n])^p}{x (d + e x^r)} \, dx$ when $p \in \mathbb{Z}^+$

Rule: Integration by parts

Basis: $\frac{1}{x (d+e x^r)} == -\frac{1}{d r} \partial_x \operatorname{Log}\left[1 + \frac{d}{e x^r}\right]$

Rule: If $p \in \mathbb{Z}^+$, then

$$\int \frac{(a + b \operatorname{Log}[c x^n])^p}{x (d + e x^r)} \, dx \longrightarrow -\frac{\operatorname{Log}\left[1 + \frac{d}{e x^r}\right] (a + b \operatorname{Log}[c x^n])^p}{d r} + \frac{b n p}{d r} \int \frac{\operatorname{Log}\left[1 + \frac{d}{e x^r}\right] (a + b \operatorname{Log}[c x^n])^{p-1}}{x} \, dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])^p_./(x_*(d_+e_.*x_^r_.)),x_Symbol] :=
  -Log[1+d/(e*x^r)]*(a+b*Log[c*x^n])^p/(d*r) +
  b*n*p/(d*r)*Int[Log[1+d/(e*x^r)]*(a+b*Log[c*x^n])^(p-1)/x,x] /;
FreeQ[{a,b,c,d,e,n,r},x] && IGtQ[p,0]
```

2: $\int \frac{x^m (a + b \operatorname{Log}[c x^n])^p}{d + e x^r} \, dx$ when $p \in \mathbb{Z}^+ \bigwedge r \in \mathbb{Z}^+ \bigwedge m + 1 \in \mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: $\frac{x^m}{d+e x^r} == \frac{x^m}{d} - \frac{e x^{m+r}}{d (d+e x^r)}$

Rule: If $p \in \mathbb{Z}^+ \bigwedge r \in \mathbb{Z}^+ \bigwedge m + 1 \in \mathbb{Z}^-$, then

$$\int \frac{x^m (a + b \operatorname{Log}[c x^n])^p}{d + e x^r} \, dx \longrightarrow \frac{1}{d} \int x^m (a + b \operatorname{Log}[c x^n])^p \, dx - \frac{e}{d} \int \frac{x^{m+r} (a + b \operatorname{Log}[c x^n])^p}{d + e x^r} \, dx$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Log[c_.*x_^n_.])^p_./(d_+e_.*x_^r_.),x_Symbol] :=
  1/d*Int[x^m*(a+b*Log[c*x^n])^p,x] -
  e/d*Int[(x^(m+r)*(a+b*Log[c*x^n])^p)/(d+e*x^r),x] /;
FreeQ[{a,b,c,d,e,m,n,r},x] && IGtQ[p,0] && IGtQ[r,0] && ILtQ[m,-1]
```

?. $\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$ when $m+q+1\in\mathbb{Z}^-\wedge p\in\mathbb{Z}^+\wedge q<-1$

1: $\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$ when $m+q+2==0\wedge p\in\mathbb{Z}^+\wedge q<-1$

Derivation: Integration by parts

- **Basis: If** $m+q+2==0$, **then** $(f\,x)^m\,(d+e\,x)^q == -\partial_x\frac{(f\,x)^{m+1}\,(d+e\,x)^{q+1}}{d\,f\,(q+1)}$
- **Basis:** $\partial_x\,(a+b\,\text{Log}[c\,x^n])^p == \frac{b\,n\,p\,(a+b\,\text{Log}[c\,x^n])^{p-1}}{x}$

Rule: If $m+q+2==0\wedge p\in\mathbb{Z}^+\wedge q<-1$, **then**

$$\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx\longrightarrow$$
$$-\frac{(f\,x)^{m+1}\,(d+e\,x)^{q+1}\,(a+b\,\text{Log}[c\,x^n])^p}{d\,f\,(q+1)}+\frac{b\,n\,p}{d\,(q+1)}\int (f\,x)^m\,(d+e\,x)^{q+1}\,(a+b\,\text{Log}[c\,x^n])^{p-1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_)^q_*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  -(f*x)^(m+1)*(d+e*x)^(q+1)*(a+b*Log[c*x^n])^p/(d*f*(q+1)) +
  b*n*p/(d*(q+1))*Int[(f*x)^m*(d+e*x)^(q+1)*(a+b*Log[c*x^n])^(p-1),x] /;
FreeQ[{a,b,c,d,e,f,m,n,q},x] && EqQ[m+q+2,0] && IGtQ[p,0] && LtQ[q,-1]
```

2. $\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$ when $m+q+2\in\mathbb{Z}^-\land\,p\in\mathbb{Z}^+\land\,q<-1\,\land\,m>0$

1: $\int x^m\,(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])\,dx$ when $m+q+2\in\mathbb{Z}^-\land\,m\in\mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{Log}[c\,x^n]) == \frac{b\,n}{x}$

Rule: If $m+q+2\in\mathbb{Z}^-\land\,m\in\mathbb{Z}^+$, **let** $u\to\int x^m\,(d+e\,x)^q\,dx$, **then**

$$\int x^m\,(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])\,dx\;\longrightarrow\;u\,(a+b\,\text{Log}[c\,x^n])-b\,n\int\frac{u}{x}\,dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_)^q_*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  With[{u=IntHide[x^m*(d+e*x)^q,x]},
  Dist[(a+b*Log[c*x^n]),u,x] - b*n*Int[SimplifyIntegrand[u/x,x],x]] /;
FreeQ[{a,b,c,d,e,n},x] && ILtQ[m+q+2,0] && IGtQ[m,0]
```

2: $\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$ when $m+q+2\in\mathbb{Z}^-\land\,p\in\mathbb{Z}^+\land\,q<-1\,\land\,m>0$

Derivation: Algebraic expansion and integration by parts

Basis: $(d+e\,x)^q == -\frac{(d+e\,x)^q\,(d\,(m+1)+e\,(m+q+2)\,x)}{d\,(q+1)}+\frac{(m+q+2)\,(d+e\,x)^{q+1}}{d\,(q+1)}$

Basis: $(f\,x)^m\,(d+e\,x)^q\,(d\,(m+1)+e\,(m+q+2)\,x) == \partial_x\,\frac{(f\,x)^{m+1}\,(d+e\,x)^{q+1}}{f}$

Basis: $\partial_x\,(a+b\,\text{Log}[c\,x^n])^p == \frac{b\,n\,p\,(a+b\,\text{Log}[c\,x^n])^{p-1}}{x}$

Rule: If $m+q+2\in\mathbb{Z}^-\land\,p\in\mathbb{Z}^+\land\,q<-1\,\land\,m>0$, **then**

$$\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$$

$$\longrightarrow\;-\frac{1}{d\,(q+1)}\int (f\,x)^m\,(d+e\,x)^q\,(d\,(m+1)+e\,(m+q+2)\,x)\,(a+b\,\text{Log}[c\,x^n])^p\,dx+\frac{m+q+2}{d\,(q+1)}\int (f\,x)^m\,(d+e\,x)^{q+1}\,(a+b\,\text{Log}[c\,x^n])^p\,dx$$

$$\longrightarrow\;-\frac{(f\,x)^{m+1}\,(d+e\,x)^{q+1}\,(a+b\,\text{Log}[c\,x^n])^p}{d\,f\,(q+1)}+\frac{b\,n\,p}{d\,(q+1)}\int (f\,x)^m\,(d+e\,x)^{q+1}\,(a+b\,\text{Log}[c\,x^n])^{p-1}\,dx\;+$$

$$\frac{m+q+2}{d\,(q+1)}\int (f\,x)^m\,(d+e\,x)^{q+1}\,(a+b\,\text{Log}[c\,x^n])^p\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_)^q_*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  -(f*x)^(m+1)*(d+e*x)^(q+1)*(a+b*Log[c*x^n])^p/(d*f*(q+1)) +
  (m+q+2)/(d*(q+1))*Int[(f*x)^m*(d+e*x)^(q+1)*(a+b*Log[c*x^n])^p,x] +
  b*n*p/(d*(q+1))*Int[(f*x)^m*(d+e*x)^(q+1)*(a+b*Log[c*x^n])^(p-1),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && ILtQ[m+q+2,0] && IGtQ[p,0] && LtQ[q,-1] && GtQ[m,0]
```

4. $\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])\,dx$ when $q+1\in\mathbb{Z}^-$

1: $\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])\,dx$ when $q+1\in\mathbb{Z}^-\bigwedge m>0$

Rule: If $q+1\in\mathbb{Z}^-\bigwedge m>0$, then

$$\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])\,dx\longrightarrow$$
$$\frac{(f\,x)^m\,(d+e\,x)^{q+1}\,(a+b\,\text{Log}[c\,x^n])}{e\,(q+1)}-\frac{f}{e\,(q+1)}\int (f\,x)^{m-1}\,(d+e\,x)^{q+1}\,(a\,m+b\,n+b\,m\,\text{Log}[c\,x^n])\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_)^q_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  (f*x)^m*(d+e*x)^(q+1)*(a+b*Log[c*x^n])/(e*(q+1)) -
  f/(e*(q+1))*Int[(f*x)^(m-1)*(d+e*x)^(q+1)*(a*m+b*n+b*m*Log[c*x^n]),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && ILtQ[q,-1] && GtQ[m,0]
```

2: $\int (f x)^m \left(d+e x^2\right)^q (a+b \operatorname{Log}[c x^n])\, dx$ when $q+1 \in \mathbb{Z}^- \wedge m \in \mathbb{Z}^-$

Rule: If $q+1 \in \mathbb{Z}^- \wedge m \in \mathbb{Z}^-$, then

$$\int (f x)^m \left(d+e x^2\right)^q (a+b \operatorname{Log}[c x^n])\, dx \longrightarrow$$
$$-\frac{(f x)^{m+1} \left(d+e x^2\right)^{q+1} (a+b \operatorname{Log}[c x^n])}{2 d f (q+1)}+\frac{1}{2 d (q+1)} \int (f x)^m \left(d+e x^2\right)^{q+1} (a (m+2 q+3)+b n+b (m+2 q+3) \operatorname{Log}[c x^n])\, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  -(f*x)^(m+1)*(d+e*x^2)^(q+1)*(a+b*Log[c*x^n])/(2*d*f*(q+1)) +
  1/(2*d*(q+1))*Int[(f*x)^m*(d+e*x^2)^(q+1)*(a*(m+2*q+3)+b*n+b*(m+2*q+3)*Log[c*x^n]),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && ILtQ[q,-1] && ILtQ[m,0]
```

5: $\int x^m \left(d+e x^2\right)^q (a+b \operatorname{Log}[c x^n])\, dx$ when $\frac{m}{2} \in \mathbb{Z} \bigwedge q-\frac{1}{2} \in \mathbb{Z} \bigwedge \neg (m+2 q<-2 \vee d>0)$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\left(d+e x^2\right)^q}{\left(1+\frac{e}{d} x^2\right)^q} = 0$

- **Rule: If $\frac{m}{2} \in \mathbb{Z} \bigwedge q-\frac{1}{2} \in \mathbb{Z} \bigwedge \neg (m+2 q<-2 \vee d>0)$, then**

$$\int x^m \left(d+e x^2\right)^q (a+b \operatorname{Log}[c x^n])\, dx \longrightarrow \frac{d^{\operatorname{IntPart}[q]} \left(d+e x^2\right)^{\operatorname{FracPart}[q]}}{\left(1+\frac{e}{d} x^2\right)^{\operatorname{FracPart}[q]}} \int x^m \left(1+\frac{e}{d} x^2\right)^q (a+b \operatorname{Log}[c x^n])\, dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^2)^q_*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  d^IntPart[q]*(d+e*x^2)^FracPart[q]/(1+e/d*x^2)^FracPart[q]*Int[x^m*(1+e/d*x^2)^q*(a+b*Log[c*x^n]),x] /;
FreeQ[{a,b,c,d,e,n},x] && IntegerQ[m/2] && IntegerQ[q-1/2] && Not[LtQ[m+2*q,-2] || GtQ[d,0]]
```

```
Int[x_^m_.*(d1_+e1_.*x_)^q_*(d2_+e2_.*x_)^q_*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  (d1+e1*x)^q*(d2+e2*x)^q/(1+e1*e2/(d1*d2)*x^2)^q*Int[x^m*(1+e1*e2/(d1*d2)*x^2)^q*(a+b*Log[c*x^n]),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && EqQ[d2*e1+d1*e2,0] && IntegerQ[m] && IntegerQ[q-1/2]
```

6. $\int \frac{(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx$ when $p\in\mathbb{Z}^+$

1. $\int \frac{(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx$ when $p\in\mathbb{Z}^+$

1: $\int \frac{(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx$ when $p\in\mathbb{Z}^+\wedge q>0$

Rule: Algebraic expansion

Basis: $\frac{(d+e\,x)^q}{x}=\frac{d\,(d+e\,x)^{q-1}}{x}+e\,(d+e\,x)^{q-1}$

Rule: If $p\in\mathbb{Z}^+\wedge q>0$, then

$$\int \frac{(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx\longrightarrow d\int \frac{(d+e\,x)^{q-1}\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx+e\int (d+e\,x)^{q-1}\,(a+b\,\text{Log}[c\,x^n])^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_./x_,x_Symbol] :=
  d*Int[(d+e*x)^(q-1)*(a+b*Log[c*x^n])^p/x,x] +
  e*Int[(d+e*x)^(q-1)*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[p,0] && GtQ[q,0] && IntegerQ[2*q]
```

2: $\int \frac{(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx$ when $p\in\mathbb{Z}^+\wedge q<-1$

Rule: Algebraic expansion

Basis: $\frac{(d+e\,x)^q}{x}=\frac{(d+e\,x)^{q+1}}{d\,x}-\frac{e\,(d+e\,x)^q}{d}$

Rule: If $p\in\mathbb{Z}^+\wedge q<-1$, then

$$\int \frac{(d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx\longrightarrow \frac{1}{d}\int \frac{(d+e\,x)^{q+1}\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx-\frac{e}{d}\int (d+e\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)^q_*(a_.+b_.*Log[c_.*x_^n_.])^p_./x_,x_Symbol] :=
  1/d*Int[(d+e*x)^(q+1)*(a+b*Log[c*x^n])^p/x,x] -
  e/d*Int[(d+e*x)^q*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[p,0] && LtQ[q,-1] && IntegerQ[2*q]
```

2: $\int \frac{(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])}{x}\,dx$ when $q-\frac{1}{2}\in\mathbb{Z}$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{Log}[c\,x^n]) == \frac{b\,n}{x}$

Rule: If $q-\frac{1}{2}\in\mathbb{Z}$, **let** $u\to\int\frac{(d+e\,x^r)^q}{x}\,dx$, **then**

$$\int \frac{(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])}{x}\,dx \longrightarrow u\,(a+b\,\text{Log}[c\,x^n])-b\,n\int\frac{u}{x}\,dx$$

Program code:

```
Int[(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.])/x_,x_Symbol] :=
  With[{u=IntHide[(d+e*x^r)^q/x,x]},
  u*(a+b*Log[c*x^n]) - b*n*Int[Dist[1/x,u,x],x]] /;
FreeQ[{a,b,c,d,e,n,r},x] && IntegerQ[q-1/2]
```

3: $\int \frac{(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx$ when $p\in\mathbb{Z}^+\wedge q+1\in\mathbb{Z}^-$

Rule: Algebraic expansion

Basis: $\frac{(d+e\,x^r)^q}{x} == \frac{(d+e\,x^r)^{q+1}}{d\,x}-\frac{e\,x^{r-1}\,(d+e\,x^r)^q}{d}$

Rule: If $p\in\mathbb{Z}^+\wedge q+1\in\mathbb{Z}^-$, **then**

$$\int \frac{(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx \longrightarrow \frac{1}{d}\int\frac{(d+e\,x^r)^{q+1}\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx-\frac{e}{d}\int x^{r-1}\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$$

Program code:

```
Int[(d_+e_.*x_^r_.)^q_*(a_.+b_.*Log[c_.*x_^n_.])^p_./x_,x_Symbol] :=
  1/d*Int[(d+e*x^r)^(q+1)*(a+b*Log[c*x^n])^p/x,x] -
  e/d*Int[x^(r-1)*(d+e*x^r)^q*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,n,r},x] && IGtQ[p,0] && ILtQ[q,-1]
```

7: $\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])\,dx$ when $m\in\mathbb{Z}\,\wedge\,2q\in\mathbb{Z}\,\wedge\,r\in\mathbb{Z}$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{Log}[c\,x^n]) == \frac{b\,n}{x}$

Note: If $m\in\mathbb{Z}\,\wedge\,q-\frac{1}{2}\in\mathbb{Z}$, then the terms of $\int x^m\,(d+e\,x)^q\,dx$ will be algebraic functions or constants times an inverse function.

Rule: If $m\in\mathbb{Z}\,\wedge\,2q\in\mathbb{Z}\,\wedge\,r\in\mathbb{Z}$, let $u\to\int (f\,x)^m\,(d+e\,x^r)^q\,dx$, then

$$\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])\,dx\;\longrightarrow\;u\,(a+b\,\text{Log}[c\,x^n])-b\,n\int\frac{u}{x}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^r)^q,x]},
  Dist[(a+b*Log[c*x^n]),u,x] - b*n*Int[SimplifyIntegrand[u/x,x],x] /;
 (EqQ[r,1] || EqQ[r,2]) && IntegerQ[m] && IntegerQ[q-1/2] || InverseFunctionFreeQ[u,x]] /;
FreeQ[{a,b,c,d,e,f,m,n,q,r},x] && IntegerQ[2*q] && (IntegerQ[m] && IntegerQ[r] || IGtQ[q,0])
```

8: $\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])\,dx$ when $q\in\mathbb{Z}\,\wedge\,(q>0\,\vee\,m\in\mathbb{Z}\wedge r\in\mathbb{Z})$

Derivation: Algebraic expansion

Rule: If $q\in\mathbb{Z}\,\wedge\,(q>0\,\vee\,m\in\mathbb{Z}\wedge r\in\mathbb{Z})$, then

$$\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])\,dx\;\longrightarrow\;\int (a+b\,\text{Log}[c\,x^n])\,\text{ExpandIntegrand}[(f\,x)^m\,(d+e\,x^r)^q,\,x]\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  With[{u=ExpandIntegrand[(a+b*Log[c*x^n]),(f*x)^m*(d+e*x^r)^q,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,f,m,n,q,r},x] && IntegerQ[q] && (GtQ[q,0] || IntegerQ[m] && IntegerQ[r])
```

9: $\int x^m (d+e x^r)^q (a+b \operatorname{Log}[c x^n])^p \, dx$ when $q \in \mathbb{Z} \bigwedge \frac{r}{n} \in \mathbb{Z} \bigwedge \frac{m+1}{n} \in \mathbb{Z} \bigwedge \left(\frac{m+1}{n} > 0 \bigvee p \in \mathbb{Z}^+\right)$

Derivation: Integration by substitution

- **Basis: If** $\frac{m+1}{n} \in \mathbb{Z}$**, then** $x^m F[x^n] = \frac{1}{n} \operatorname{Subst}\left[x^{\frac{m+1}{n}-1} F[x],\, x,\, x^n\right] \partial_x x^n$
- **Rule: If** $q \in \mathbb{Z} \bigwedge \frac{r}{n} \in \mathbb{Z} \bigwedge \frac{m+1}{n} \in \mathbb{Z} \bigwedge \left(\frac{m+1}{n} > 0 \bigvee p \in \mathbb{Z}^+\right)$**, then**

$$\int x^m (d+e x^r)^q (a+b \operatorname{Log}[c x^n])^p \, dx \longrightarrow \frac{1}{n} \operatorname{Subst}\left[\int x^{\frac{m+1}{n}-1} \left(d+e x^{\frac{r}{n}}\right)^q (a+b \operatorname{Log}[c x])^p \, dx,\, x,\, x^n\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(d+e*x^(r/n))^q*(a+b*Log[c*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,e,m,n,p,q,r},x] && IntegerQ[q] && IntegerQ[r/n] && IntegerQ[Simplify[(m+1)/n]] && (GtQ[(m+1)/n,0] || IGtQ[p,0])
```

10: $\int (f x)^m (d+e x^r)^q (a+b \operatorname{Log}[c x^n])^p \, dx$ when $q \in \mathbb{Z} \wedge (q > 0 \vee p \in \mathbb{Z}^+ \wedge m \in \mathbb{Z} \wedge r \in \mathbb{Z})$

Derivation: Algebraic expansion

Rule: If $q \in \mathbb{Z} \wedge (q > 0 \vee p \in \mathbb{Z}^+ \wedge m \in \mathbb{Z} \wedge r \in \mathbb{Z})$**, then**

$$\int (f x)^m (d+e x^r)^q (a+b \operatorname{Log}[c x^n])^p \, dx \longrightarrow \int (a+b \operatorname{Log}[c x^n])^p \operatorname{ExpandIntegrand}[(f x)^m (d+e x^r)^q,\, x] \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  With[{u=ExpandIntegrand[(a+b*Log[c*x^n])^p,(f*x)^m*(d+e*x^r)^q,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q,r},x] && IntegerQ[q] && (GtQ[q,0] || IGtQ[p,0] && IntegerQ[m] && IntegerQ[r])
```

U: $\int (f x)^m (d+e x^r)^q (a+b \operatorname{Log}[c x^n])^p \, dx$

Rule:

$$\int (f x)^m (d+e x^r)^q (a+b \operatorname{Log}[c x^n])^p \, dx \longrightarrow \int (f x)^m (d+e x^r)^q (a+b \operatorname{Log}[c x^n])^p \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Unintegrable[(f*x)^m*(d+e*x^r)^q*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q,r},x]
```

N: $\int (f x)^m u^q (a+b \operatorname{Log}[c x^n])^p \, dx$ when $u == d+e x^r$

Derivation: Algebraic normalization

Rule: If $u == d+e x^r$, then

$$\int (f x)^m u^q (a+b \operatorname{Log}[c x^n])^p \, dx \longrightarrow \int (f x)^m (d+e x^r)^q (a+b \operatorname{Log}[c x^n])^p \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*u_^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Int[(f*x)^m*ExpandToSum[u,x]^q*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,f,m,n,p,q},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

Rules for integrands of the form $(f+g x)^m (d+e x)^q (a+b \operatorname{Log}[c x^n])^p$

1: $\int (f+g x)^m (d+e x)^q (a+b \operatorname{Log}[c x^n])^p \, dx$ when $e f-d g \neq 0 \land m+q+2 == 0 \land p \in \mathbb{Z}^+ \land q < -1$

- **Derivation: Integration by parts**
- **Basis: If $m+q+2 == 0$, then $(f+g x)^m (d+e x)^q == \partial_x \frac{(f+g x)^{m+1} (d+e x)^{q+1}}{(q+1) (e f-d g)}$**
- **Basis: $\partial_x (a+b \operatorname{Log}[c x^n])^p == \frac{b n p (a+b \operatorname{Log}[c x^n])^{p-1}}{x}$**
- **Rule: If $e f-d g \neq 0 \land m+q+2 == 0 \land p \in \mathbb{Z}^+ \land q < -1$, then**

$$\int (f+g\,x)^m\,(d+e\,x)^q\,(a+b\,\mathrm{Log}[c\,x^n])^p\,dx\ \longrightarrow$$
$$\frac{(f+g\,x)^{m+1}\,(d+e\,x)^{q+1}\,(a+b\,\mathrm{Log}[c\,x^n])^p}{(q+1)\,(e\,f-d\,g)}-\frac{b\,n\,p}{(q+1)\,(e\,f-d\,g)}\int\frac{(f+g\,x)^{m+1}\,(d+e\,x)^{q+1}\,(a+b\,\mathrm{Log}[c\,x^n])^{p-1}}{x}\,dx$$

- **Program code:**

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_)^q_*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  (f+g*x)^(m+1)*(d+e*x)^(q+1)*(a+b*Log[c*x^n])^p/((q+1)*(e*f-d*g)) -
  b*n*p/((q+1)*(e*f-d*g))*Int[(f+g*x)^(m+1)*(d+e*x)^(q+1)*(a+b*Log[c*x^n])^(p-1)/x,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,q},x] && NeQ[e*f-d*g,0] && EqQ[m+q+2,0] && IGtQ[p,0] && LtQ[q,-1]
```