Rules for integrands of the form $(c + d\,x)^m\,\text{Trig}[a + b\,x]^n\,\text{Trig}[a + b\,x]^p$

1. $\int (c + d\,x)^m\,\text{Trig}[a + b\,x]^n\,\text{Trig}[a + b\,x]^p\,dx$

   1. $\int (c + d\,x)^m\,\text{Sin}[a + b\,x]^n\,\text{Cos}[a + b\,x]^p\,dx$

      1: $\int (c + d\,x)^m\,\text{Sin}[a + b\,x]^n\,\text{Cos}[a + b\,x]\,dx$ when $m \in \mathbb{Z}^+ \land n \neq -1$

- **Derivation: Integration by parts**
- **Basis:** $\text{Sin}[a + b\,x]^n\,\text{Cos}[a + b\,x] == \partial_x \frac{\text{Sin}[a+b\,x]^{n+1}}{b\,(n+1)}$
- **Rule: If** $m \in \mathbb{Z}^+ \land n \neq -1$, **then**

$$\int (c + d\,x)^m\,\text{Sin}[a + b\,x]^n\,\text{Cos}[a + b\,x]\,dx \longrightarrow \frac{(c + d\,x)^m\,\text{Sin}[a + b\,x]^{n+1}}{b\,(n + 1)} - \frac{d\,m}{b\,(n + 1)} \int (c + d\,x)^{m-1}\,\text{Sin}[a + b\,x]^{n+1}\,dx$$

- **Program code:**

```
Int[(c_.+d_.*x_)^m_.*Sin[a_.+b_.*x_]^n_.*Cos[a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^m*Sin[a+b*x]^(n+1)/(b*(n+1)) -
  d*m/(b*(n+1))*Int[(c+d*x)^(m-1)*Sin[a+b*x]^(n+1),x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

```
Int[(c_.+d_.*x_)^m_.*Sin[a_.+b_.*x_]*Cos[a_.+b_.*x_]^n_.,x_Symbol] :=
  -(c+d*x)^m*Cos[a+b*x]^(n+1)/(b*(n+1)) +
  d*m/(b*(n+1))*Int[(c+d*x)^(m-1)*Cos[a+b*x]^(n+1),x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

2: $\int (c + d\,x)^m\,\text{Sin}[a + b\,x]^n\,\text{Cos}[a + b\,x]^p\,dx$ when $(n \mid p) \in \mathbb{Z}^+$

- **Derivation: Algebraic expansion**
- **Rule: If** $(n \mid p) \in \mathbb{Z}^+$, **then**

$$\int (c + d\,x)^m\,\text{Sin}[a + b\,x]^n\,\text{Cos}[a + b\,x]^p\,dx \longrightarrow \int (c + d\,x)^m\,\text{TrigReduce}[\text{Sin}[a + b\,x]^n\,\text{Cos}[a + b\,x]^p]\,dx$$

- **Program code:**

```
Int[(c_.+d_.*x_)^m_.*Sin[a_.+b_.*x_]^n_.*Cos[a_.+b_.*x_]^p_.,x_Symbol] :=
  Int[ExpandTrigReduce[(c+d*x)^m,Sin[a+b*x]^n*Cos[a+b*x]^p,x],x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[n,0] && IGtQ[p,0]
```

2: $\int (c+d\,x)^m \operatorname{Sin}[a+b\,x]^n \operatorname{Tan}[a+b\,x]^p \,dx$ when $(n \mid p) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\operatorname{Sin}[z]^2 \operatorname{Tan}[z]^2 = -\operatorname{Sin}[z]^2 + \operatorname{Tan}[z]^2$

Rule: If $(n \mid p) \in \mathbb{Z}^+$**, then**

$$\int (c+d\,x)^m \operatorname{Sin}[a+b\,x]^n \operatorname{Tan}[a+b\,x]^p \,dx \longrightarrow$$
$$-\int (c+d\,x)^m \operatorname{Sin}[a+b\,x]^n \operatorname{Tan}[a+b\,x]^{p-2} \,dx + \int (c+d\,x)^m \operatorname{Sin}[a+b\,x]^{n-2} \operatorname{Tan}[a+b\,x]^p \,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Sin[a_.+b_.*x_]^n_.*Tan[a_.+b_.*x_]^p_.,x_Symbol] :=
  -Int[(c+d*x)^m*Sin[a+b*x]^n*Tan[a+b*x]^(p-2),x] + Int[(c+d*x)^m*Sin[a+b*x]^(n-2)*Tan[a+b*x]^p,x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[n,0] && IGtQ[p,0]
```

```
Int[(c_.+d_.*x_)^m_.*Cos[a_.+b_.*x_]^n_.*Cot[a_.+b_.*x_]^p_.,x_Symbol] :=
  -Int[(c+d*x)^m*Cos[a+b*x]^n*Cot[a+b*x]^(p-2),x] + Int[(c+d*x)^m*Cos[a+b*x]^(n-2)*Cot[a+b*x]^p,x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[n,0] && IGtQ[p,0]
```

3. $\int (c+d\,x)^m \operatorname{Sec}[a+b\,x]^n \operatorname{Tan}[a+b\,x]^p \, dx$

1: $\int (c+d\,x)^m \operatorname{Sec}[a+b\,x]^n \operatorname{Tan}[a+b\,x] \, dx$ when $m>0$

Derivation: Integration by parts

Basis: $\operatorname{Sec}[a+b\,x]^n \operatorname{Tan}[a+b\,x] = \partial_x \frac{\operatorname{Sec}[a+b\,x]^n}{b\,n}$

Note: Dummy exponent $p === 1$ required in program code so InputForm of integrand is recognized.

Rule: If $m>0$, then

$$\int (c+d\,x)^m \operatorname{Sec}[a+b\,x]^n \operatorname{Tan}[a+b\,x] \, dx \longrightarrow \frac{(c+d\,x)^m \operatorname{Sec}[a+b\,x]^n}{b\,n} - \frac{d\,m}{b\,n} \int (c+d\,x)^{m-1} \operatorname{Sec}[a+b\,x]^n \, dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Sec[a_.+b_.*x_]^n_.*Tan[a_.+b_.*x_]^p_.,x_Symbol] :=
  (c+d*x)^m*Sec[a+b*x]^n/(b*n) -
  d*m/(b*n)*Int[(c+d*x)^(m-1)*Sec[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x] && EqQ[p,1] && GtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_.*Csc[a_.+b_.*x_]^n_.*Cot[a_.+b_.*x_]^p_.,x_Symbol] :=
  -(c+d*x)^m*Csc[a+b*x]^n/(b*n) +
  d*m/(b*n)*Int[(c+d*x)^(m-1)*Csc[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x] && EqQ[p,1] && GtQ[m,0]
```

2: $\int (c+d\,x)^m \operatorname{Sec}[a+b\,x]^2 \operatorname{Tan}[a+b\,x]^n \, dx$ when $m \in \mathbb{Z}^+ \wedge n \neq -1$

Derivation: Integration by parts

- **Basis:** $\operatorname{Sec}[a+b\,x]^2 \operatorname{Tan}[a+b\,x]^n == \partial_x \frac{\operatorname{Tan}[a+b\,x]^{n+1}}{b\,(n+1)}$

Rule: If $m \in \mathbb{Z}^+ \wedge n \neq -1$**, then**

$$\int (c+d\,x)^m \operatorname{Sec}[a+b\,x]^2 \operatorname{Tan}[a+b\,x]^n \, dx \longrightarrow \frac{(c+d\,x)^m \operatorname{Tan}[a+b\,x]^{n+1}}{b\,(n+1)} - \frac{d\,m}{b\,(n+1)} \int (c+d\,x)^{m-1} \operatorname{Tan}[a+b\,x]^{n+1} \, dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Sec[a_.+b_.*x_]^2*Tan[a_.+b_.*x_]^n_.,x_Symbol] :=
  (c+d*x)^m*Tan[a+b*x]^(n+1)/(b*(n+1)) -
  d*m/(b*(n +1))*Int[(c+d*x)^(m-1)*Tan[a+b*x]^(n+1),x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

```
Int[(c_.+d_.*x_)^m_.*Csc[a_.+b_.*x_]^2*Cot[a_.+b_.*x_]^n_.,x_Symbol] :=
  -(c+d*x)^m*Cot[a+b*x]^(n+1)/(b*(n+1)) +
  d*m/(b*(n +1))*Int[(c+d*x)^(m-1)*Cot[a+b*x]^(n+1),x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

3: $\int (c+d\,x)^m \operatorname{Sec}[a+b\,x]^n \operatorname{Tan}[a+b\,x]^p \, dx$ when $\frac{p}{2} \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\operatorname{Tan}[z]^2 == -1 + \operatorname{Sec}[z]^2$

Rule: If $\frac{p}{2} \in \mathbb{Z}^+$**, then**

$$\int (c+d\,x)^m \operatorname{Sec}[a+b\,x]^n \operatorname{Tan}[a+b\,x]^p \, dx \longrightarrow$$
$$-\int (c+d\,x)^m \operatorname{Sec}[a+b\,x]^n \operatorname{Tan}[a+b\,x]^{p-2} \, dx + \int (c+d\,x)^m \operatorname{Sec}[a+b\,x]^{n+2} \operatorname{Tan}[a+b\,x]^{p-2} \, dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Sec[a_.+b_.*x_]*Tan[a_.+b_.*x_]^p_,x_Symbol] :=
  -Int[(c+d*x)^m*Sec[a+b*x]*Tan[a+b*x]^(p-2),x] + Int[(c+d*x)^m*Sec[a+b*x]^3*Tan[a+b*x]^(p-2),x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[p/2,0]
```

```
Int[(c_.+d_.*x_)^m_.*Sec[a_.+b_.*x_]^n_.*Tan[a_.+b_.*x_]^p_,x_Symbol] :=
  -Int[(c+d*x)^m*Sec[a+b*x]^n*Tan[a+b*x]^(p-2),x] + Int[(c+d*x)^m*Sec[a+b*x]^(n+2)*Tan[a+b*x]^(p-2),x] /;
FreeQ[{a,b,c,d,m,n},x] && IGtQ[p/2,0]
```

```
Int[(c_.+d_.*x_)^m_.*Csc[a_.+b_.*x_]*Cot[a_.+b_.*x_]^p_,x_Symbol] :=
  -Int[(c+d*x)^m*Csc[a+b*x]*Cot[a+b*x]^(p-2),x] + Int[(c+d*x)^m*Csc[a+b*x]^3*Cot[a+b*x]^(p-2),x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[p/2,0]
```

```
Int[(c_.+d_.*x_)^m_.*Csc[a_.+b_.*x_]^n_.*Cot[a_.+b_.*x_]^p_,x_Symbol] :=
  -Int[(c+d*x)^m*Csc[a+b*x]^n*Cot[a+b*x]^(p-2),x] + Int[(c+d*x)^m*Csc[a+b*x]^(n+2)*Cot[a+b*x]^(p-2),x] /;
FreeQ[{a,b,c,d,m,n},x] && IGtQ[p/2,0]
```

4: $\int (c+d\,x)^m \operatorname{Sec}[a+b\,x]^n \operatorname{Tan}[a+b\,x]^p \,dx$ when $m\in\mathbb{Z}^+ \bigwedge \left(\frac{n}{2}\in\mathbb{Z} \bigvee \frac{p-1}{2}\in\mathbb{Z}\right)$

Derivation: Integration by parts

- **Rule: If** $m\in\mathbb{Z}^+ \bigwedge \left(\frac{n}{2}\in\mathbb{Z} \bigvee \frac{p-1}{2}\in\mathbb{Z}\right)$, **let** $u = \int \operatorname{Sec}[a+b\,x]^n \operatorname{Tan}[a+b\,x]^p \,dx$, **then**

$$\int (c+d\,x)^m \operatorname{Sec}[a+b\,x]^n \operatorname{Tan}[a+b\,x]^p \,dx \longrightarrow u\,(c+d\,x)^m - d\,m \int u\,(c+d\,x)^{m-1}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Sec[a_.+b_.*x_]^n_.*Tan[a_.+b_.*x_]^p_.,x_Symbol] :=
  Module[{u=IntHide[Sec[a+b*x]^n*Tan[a+b*x]^p,x]},
  Dist[(c+d*x)^m,u,x] - d*m*Int[(c+d*x)^(m-1)*u,x]] /;
FreeQ[{a,b,c,d,n,p},x] && IGtQ[m,0] && (IntegerQ[n/2] || IntegerQ[(p-1)/2])
```

```
Int[(c_.+d_.*x_)^m_.*Csc[a_.+b_.*x_]^n_.*Cot[a_.+b_.*x_]^p_.,x_Symbol] :=
  Module[{u=IntHide[Csc[a+b*x]^n*Cot[a+b*x]^p,x]},
  Dist[(c+d*x)^m,u,x] - d*m*Int[(c+d*x)^(m-1)*u,x]] /;
FreeQ[{a,b,c,d,n,p},x] && IGtQ[m,0] && (IntegerQ[n/2] || IntegerQ[(p-1)/2])
```

4. $\int (c+d\,x)^m \operatorname{Sec}[a+b\,x]^p \operatorname{Csc}[a+b\,x]^n \,dx$

1: $\int (c+d\,x)^m \operatorname{Csc}[a+b\,x]^n \operatorname{Sec}[a+b\,x]^n \,dx$ when $n \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: $\operatorname{Csc}[z] \operatorname{Sec}[z] == 2 \operatorname{Csc}[2\,z]$

Rule: If $n \in \mathbb{Z}$, **then**

$$\int (c+d\,x)^m \operatorname{Csc}[a+b\,x]^n \operatorname{Sec}[a+b\,x]^n \,dx \longrightarrow 2^n \int (c+d\,x)^m \operatorname{Csc}[2\,a+2\,b\,x]^n \,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Csc[a_.+b_.*x_]^n_.*Sec[a_.+b_.*x_]^n_., x_Symbol] :=
  2^n*Int[(c+d*x)^m*Csc[2*a+2*b*x]^n,x] /;
FreeQ[{a,b,c,d,m},x] && IntegerQ[n] && RationalQ[m]
```

2: $\int (c+d\,x)^m \operatorname{Csc}[a+b\,x]^n \operatorname{Sec}[a+b\,x]^p \,dx$ when $(n \mid p) \in \mathbb{Z} \bigwedge m>0 \bigwedge n \neq p$

Derivation: Integration by parts

Rule: If $(n \mid p) \in \mathbb{Z} \bigwedge m>0 \bigwedge n \neq p$, **let** $u = \int \operatorname{Csc}[a+b\,x]^n \operatorname{Sec}[a+b\,x]^p \,dx$, **then**

$$\int (c+d\,x)^m \operatorname{Csc}[a+b\,x]^n \operatorname{Sec}[a+b\,x]^p \,dx \longrightarrow (c+d\,x)^m\,u - d\,m \int (c+d\,x)^{m-1}\,u \,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Csc[a_.+b_.*x_]^n_.*Sec[a_.+b_.*x_]^p_., x_Symbol] :=
  Module[{u=IntHide[Csc[a+b*x]^n*Sec[a+b*x]^p,x]},
  Dist[(c+d*x)^m,u,x] - d*m*Int[(c+d*x)^(m-1)*u,x]] /;
FreeQ[{a,b,c,d},x] && IntegersQ[n,p] && GtQ[m,0] && NeQ[n,p]
```

5: $\int u^m \text{Trig}[v]^n \text{Trig}[w]^p \, dx$ when $u == c + d x \land v == w == a + b x$

Derivation: Algebraic normalization

Rule: If $u == c + d x \land v == w == a + b x$, **then**

$$\int u^m \text{Trig}[v]^n \text{Trig}[w]^p \, dx \longrightarrow \int (c + d x)^m \text{Trig}[a + b x]^n \text{Trig}[a + b x]^p \, dx$$

Program code:

```
Int[u_^m_.*F_[v_]^n_.*G_[w_]^p_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*F[ExpandToSum[v,x]]^n*G[ExpandToSum[v,x]]^p,x] /;
FreeQ[{m,n,p},x] && TrigQ[F] && TrigQ[G] && EqQ[v,w] && LinearQ[{u,v,w},x] && Not[LinearMatchQ[{u,v,w},x]]
```

2: $\int (e + f x)^m \text{Cos}[c + d x] (a + b \text{Sin}[c + d x])^n \, dx$ when $m \in \mathbb{Z}^+ \land n \neq -1$

Derivation: Integration by parts

Basis: $\text{Cos}[c + d x] (a + b \text{Sin}[c + d x])^n == \partial_x \frac{(a+b \text{Sin}[c+d x])^{n+1}}{b d (n+1)}$

Rule: If $m \in \mathbb{Z}^+ \land n \neq -1$, **then**

$$\int (e + f x)^m \text{Cos}[c + d x] (a + b \text{Sin}[c + d x])^n \, dx \longrightarrow \frac{(e + f x)^m (a + b \text{Sin}[c + d x])^{n+1}}{b d (n + 1)} - \frac{f m}{b d (n + 1)} \int (e + f x)^{m-1} (a + b \text{Sin}[c + d x])^{n+1} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Cos[c_.+d_.*x_]*(a_+b_.*Sin[c_.+d_.*x_])^n_.,x_Symbol] :=
  (e+f*x)^m*(a+b*Sin[c+d*x])^(n+1)/(b*d*(n+1)) -
  f*m/(b*d*(n+1))*Int[(e+f*x)^(m-1)*(a+b*Sin[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

```
Int[(e_.+f_.*x_)^m_.*Sin[c_.+d_.*x_]*(a_+b_.*Cos[c_.+d_.*x_])^n_.,x_Symbol] :=
  -(e+f*x)^m*(a+b*Cos[c+d*x])^(n+1)/(b*d*(n+1)) +
  f*m/(b*d*(n+1))*Int[(e+f*x)^(m-1)*(a+b*Cos[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

3: $\int (e+f x)^m \operatorname{Sec}[c+d x]^2 (a+b \operatorname{Tan}[c+d x])^n \, dx$ when $m \in \mathbb{Z}^+ \wedge n \neq -1$

Derivation: Integration by parts

- **Basis:** $\operatorname{Sec}[c+d x]^2 (a+b \operatorname{Tan}[c+d x])^n = \partial_x \frac{(a+b \operatorname{Tan}[c+d x])^{n+1}}{b d (n+1)}$

Rule: If $m \in \mathbb{Z}^+ \wedge n \neq -1$**, then**

$$\int (e+f x)^m \operatorname{Sec}[c+d x]^2 (a+b \operatorname{Tan}[c+d x])^n \, dx \to \frac{(e+f x)^m (a+b \operatorname{Tan}[c+d x])^{n+1}}{b d (n+1)} - \frac{f m}{b d (n+1)} \int (e+f x)^{m-1} (a+b \operatorname{Tan}[c+d x])^{n+1} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sec[c_.+d_.*x_]^2*(a_+b_.*Tan[c_.+d_.*x_])^n_.,x_Symbol] :=
  (e+f*x)^m*(a+b*Tan[c+d*x])^(n+1)/(b*d*(n+1)) -
  f*m/(b*d*(n+1))*Int[(e+f*x)^(m-1)*(a+b*Tan[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

```
Int[(e_.+f_.*x_)^m_.*Csc[c_.+d_.*x_]^2*(a_+b_.*Cot[c_.+d_.*x_])^n_.,x_Symbol] :=
  -(e+f*x)^m*(a+b*Cot[c+d*x])^(n+1)/(b*d*(n+1)) +
  f*m/(b*d*(n+1))*Int[(e+f*x)^(m-1)*(a+b*Cot[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

4: $\int (e+f x)^m \operatorname{Sec}[c+d x] \operatorname{Tan}[c+d x] (a+b \operatorname{Sec}[c+d x])^n \, dx$ when $m \in \mathbb{Z}^+ \wedge n \neq -1$

Derivation: Integration by parts

- **Basis:** $\operatorname{Sec}[c+d x] \operatorname{Tan}[c+d x] (a+b \operatorname{Sec}[c+d x])^n = \partial_x \frac{(a+b \operatorname{Sec}[c+d x])^{n+1}}{b d (n+1)}$

Rule: If $m \in \mathbb{Z}^+ \wedge n \neq -1$**, then**

$$\int (e+f x)^m \operatorname{Sec}[c+d x] \operatorname{Tan}[c+d x] (a+b \operatorname{Sec}[c+d x])^n \, dx \to \frac{(e+f x)^m (a+b \operatorname{Sec}[c+d x])^{n+1}}{b d (n+1)} - \frac{f m}{b d (n+1)} \int (e+f x)^{m-1} (a+b \operatorname{Sec}[c+d x])^{n+1} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sec[c_.+d_.*x_]*Tan[c_.+d_.*x_]*(a_+b_.*Sec[c_.+d_.*x_])^n_.,x_Symbol] :=
  (e+f*x)^m*(a+b*Sec[c+d*x])^(n+1)/(b*d*(n+1)) -
  f*m/(b*d*(n+1))*Int[(e+f*x)^(m-1)*(a+b*Sec[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

```
Int[(e_.+f_.*x_)^m_.*Csc[c_.+d_.*x_]*Cot[c_.+d_.*x_]*(a_+b_.*Csc[c_.+d_.*x_])^n_.,x_Symbol] :=
  -(e+f*x)^m*(a+b*Csc[c+d*x])^(n+1)/(b*d*(n+1)) +
  f*m/(b*d*(n+1))*Int[(e+f*x)^(m-1)*(a+b*Csc[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

5: $\int (e+f\,x)^m \operatorname{Sin}[a+b\,x]^p \operatorname{Sin}[c+d\,x]^q \,dx$ when $(p \mid q) \in \mathbb{Z}^+ \land m \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $(p \mid q) \in \mathbb{Z}^+ \land m \in \mathbb{Z}$, then

$$\int (e+f\,x)^m \operatorname{Sin}[a+b\,x]^p \operatorname{Cos}[c+d\,x]^q \,dx \longrightarrow \int (e+f\,x)^m \operatorname{TrigReduce}[\operatorname{Sin}[a+b\,x]^p \operatorname{Cos}[c+d\,x]^q] \,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sin[a_.+b_.*x_]^p_.*Sin[c_.+d_.*x_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[(e+f*x)^m,Sin[a+b*x]^p*Sin[c+d*x]^q,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && IGtQ[q,0] && IntegerQ[m]
```

```
Int[(e_.+f_.*x_)^m_.*Cos[a_.+b_.*x_]^p_.*Cos[c_.+d_.*x_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[(e+f*x)^m,Cos[a+b*x]^p*Cos[c+d*x]^q,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && IGtQ[q,0] && IntegerQ[m]
```

6: $\int (e+f\,x)^m \operatorname{Sin}[a+b\,x]^p \operatorname{Cos}[c+d\,x]^q \,dx$ when $(p \mid q) \in \mathbb{Z}^+$

- **Derivation: Algebraic expansion**

- **Rule: If $(p \mid q) \in \mathbb{Z}^+$, then**

$$\int (e+f\,x)^m \operatorname{Sin}[a+b\,x]^p \operatorname{Cos}[c+d\,x]^q \,dx \longrightarrow \int (e+f\,x)^m \operatorname{TrigReduce}[\operatorname{Sin}[a+b\,x]^p \operatorname{Cos}[c+d\,x]^q] \,dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*Sin[a_.+b_.*x_]^p_.*Cos[c_.+d_.*x_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[(e+f*x)^m,Sin[a+b*x]^p*Cos[c+d*x]^q,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[p,0] && IGtQ[q,0]
```

7: $\int (e+f x)^m \operatorname{Sin}[a+b x]^p \operatorname{Sec}[c+d x]^q \, dx$ when $(p \mid q) \in \mathbb{Z}^+ \bigwedge b c-a d == 0 \bigwedge \frac{b}{d}-1 \in \mathbb{Z}^+$

- **Derivation: Algebraic expansion**
- **Rule: If $(p \mid q) \in \mathbb{Z}^+ \bigwedge b c-a d == 0 \bigwedge \frac{b}{d}-1 \in \mathbb{Z}^+$, then**

$$\int (e+f x)^m \operatorname{Sin}[a+b x]^p \operatorname{Sec}[c+d x]^q \, dx \longrightarrow \int (e+f x)^m \operatorname{TrigExpand}\left[\operatorname{Sin}[a+b x]^p \operatorname{Cos}[c+d x]^q\right] dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*F_[a_.+b_.*x_]^p_.*G_[c_.+d_.*x_]^q_.,x_Symbol] :=
  Int[ExpandTrigExpand[(e+f*x)^m*G[c+d*x]^q,F,c+d*x,p,b/d,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && MemberQ[{Sin,Cos},F] && MemberQ[{Sec,Csc},G] && IGtQ[p,0] && IGtQ[q,0] && EqQ[b*c-a*d,0] && IGtQ[b/d
```