Rules for integrands of the form $(a + b \operatorname{Sin}[e + f x])^m \left(A + B \operatorname{Sin}[e + f x] + C \operatorname{Sin}[e + f x]^2\right)$

1: $\int (b \operatorname{Sin}[e + f x])^m \left(B \operatorname{Sin}[e + f x] + C \operatorname{Sin}[e + f x]^2\right) dx$

- **Derivation: Algebraic simplification**
- **Rule:**

$$\int (b \operatorname{Sin}[e + f x])^m \left(B \operatorname{Sin}[e + f x] + C \operatorname{Sin}[e + f x]^2\right) dx \longrightarrow \frac{1}{b} \int (b \operatorname{Sin}[e + f x])^{m+1} (B + C \operatorname{Sin}[e + f x]) dx$$

- **Program code:**

```
Int[(b_.*sin[e_.+f_.*x_])^m_.*(B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  1/b*Int[(b*Sin[e+f*x])^(m+1)*(B+C*Sin[e+f*x]),x] /;
FreeQ[{b,e,f,B,C,m},x]
```

2. $\int (b \operatorname{Sin}[e + f x])^m \left(A + C \operatorname{Sin}[e + f x]^2\right) dx$

1: $\int (b \operatorname{Sin}[e + f x])^m \left(A + C \operatorname{Sin}[e + f x]^2\right) dx$ when $A (m + 2) + C (m + 1) == 0$

- **Derivation: Nondegenerate sine recurrence 1a with** $n \to 0, \ p \to 0$
- **Rule: If** $A (m + 2) + C (m + 1) == 0$**, then**

$$\int (b \operatorname{Sin}[e + f x])^m \left(A + C \operatorname{Sin}[e + f x]^2\right) dx \longrightarrow \frac{A \operatorname{Cos}[e + f x] (b \operatorname{Sin}[e + f x])^{m+1}}{b f (m + 1)}$$

- **Program code:**

```
Int[(b_.*sin[e_.+f_.*x_])^m_.*(A_+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  A*Cos[e+f*x]*(b*Sin[e+f*x])^(m+1)/(b*f*(m+1)) /;
FreeQ[{b,e,f,A,C,m},x] && EqQ[A*(m+2)+C*(m+1),0]
```

2: $\int (b \operatorname{Sin}[e + f x])^m \left(A + C \operatorname{Sin}[e + f x]^2\right) dx$ when $m < -1$

- **Derivation: Nondegenerate sine recurrence 1a with** $n \to 0, \ p \to 0$
- **Rule: If** $m < -1$**, then**

$$\int (b \operatorname{Sin}[e + f x])^m \left(A + C \operatorname{Sin}[e + f x]^2\right) dx \longrightarrow$$

$$\frac{A\,\text{Cos}[e+f\,x]\,(b\,\text{Sin}[e+f\,x])^{m+1}}{b\,f\,(m+1)}+\frac{A\,(m+2)+C\,(m+1)}{b^2\,(m+1)}\int(b\,\text{Sin}[e+f\,x])^{m+2}\,dx$$

Program code:

```
Int[(b_.*sin[e_.+f_.*x_])^m_*(A_+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  A*Cos[e+f*x]*(b*Sin[e+f*x])^(m+1)/(b*f*(m+1)) + (A*(m+2)+C*(m+1))/(b^2*(m+1))*Int[(b*Sin[e+f*x])^(m+2),x] /;
FreeQ[{b,e,f,A,C},x] && LtQ[m,-1]
```

3. $\int(b\,\text{Sin}[e+f\,x])^m\left(A+C\,\text{Sin}[e+f\,x]^2\right)dx$ when $m\nless -1$

1: $\int\text{Sin}[e+f\,x]^m\left(A+C\,\text{Sin}[e+f\,x]^2\right)dx$ when $\frac{m+1}{2}\in\mathbb{Z}^+$

Derivation: Algebraic expansion and integration by substitution

Basis: $\text{Sin}[z]^2 == 1-\text{Cos}[z]^2$

Basis: If $\frac{m+1}{2}\in\mathbb{Z}$**, then** $\text{Sin}[e+f\,x]^m == -\frac{1}{f}\,\text{Subst}\left[\left(1-x^2\right)^{\frac{m-1}{2}},\,x,\,\text{Cos}[e+f\,x]\right]\partial_x\text{Cos}[e+f\,x]$

Rule: If $\frac{m+1}{2}\in\mathbb{Z}^+$**, then**

$$\int\text{Sin}[e+f\,x]^m\left(A+C\,\text{Sin}[e+f\,x]^2\right)dx\longrightarrow\int\text{Sin}[e+f\,x]^m\left(A+C-C\,\text{Cos}[e+f\,x]^2\right)dx$$

$$\longrightarrow -\frac{1}{f}\,\text{Subst}\left[\int\left(1-x^2\right)^{\frac{m-1}{2}}\left(A+C-C\,x^2\right)dx,\,x,\,\text{Cos}[e+f\,x]\right]$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_.*(A_+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -1/f*Subst[Int[(1-x^2)^((m-1)/2)*(A+C-C*x^2),x],x,Cos[e+f*x]] /;
FreeQ[{e,f,A,C},x] && IGtQ[(m+1)/2,0]
```

2: $\int(b\,\text{Sin}[e+f\,x])^m\left(A+C\,\text{Sin}[e+f\,x]^2\right)dx$ when $m\nless -1$

Derivation: Nondegenerate sine recurrence 1b with $m\to 0,\ p\to 0$

Rule: If $m\nless -1$**, then**

$$\int(b\,\text{Sin}[e+f\,x])^m\left(A+C\,\text{Sin}[e+f\,x]^2\right)dx\longrightarrow$$

$$-\frac{C\,\mathrm{Cos}[e+f\,x]\,(b\,\mathrm{Sin}[e+f\,x])^{m+1}}{b\,f\,(m+2)}+\frac{A\,(m+2)+C\,(m+1)}{m+2}\int(b\,\mathrm{Sin}[e+f\,x])^m\,dx$$

Program code:

```
Int[(b_.*sin[e_.+f_.*x_])^m_.*(A_+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cos[e+f*x]*(b*Sin[e+f*x])^(m+1)/(b*f*(m+2)) + (A*(m+2)+C*(m+1))/(m+2)*Int[(b*Sin[e+f*x])^m,x] /;
FreeQ[{b,e,f,A,C,m},x] && Not[LtQ[m,-1]]
```

3: $\int(a+b\,\mathrm{Sin}[e+f\,x])^m\,\left(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2\right)\,dx$ when $A\,b^2-a\,b\,B+a^2\,C==0$

Derivation: Algebraic simplification

Basis: If $A\,b^2-a\,b\,B+a^2\,C==0$, **then** $A+B\,z+C\,z^2==\frac{1}{b^2}\,(a+b\,z)\,(b\,B-a\,C+b\,C\,z)$

Rule: If $a^2-b^2\neq0\,\bigwedge\,A\,b^2-a\,b\,B+a^2\,C==0$, **then**

$$\int(a+b\,\mathrm{Sin}[e+f\,x])^m\,\left(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2\right)\,dx\,\longrightarrow\,\frac{1}{b^2}\int(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(b\,B-a\,C+b\,C\,\mathrm{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  1/b^2*Int[(a+b*Sin[e+f*x])^(m+1)*Simp[b*B-a*C+b*C*Sin[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C,m},x] && EqQ[A*b^2-a*b*B+a^2*C,0]
```

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  C/b^2*Int[(a+b*Sin[e+f*x])^(m+1)*Simp[-a+b*Sin[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C,m},x] && EqQ[A*b^2+a^2*C,0]
```

4: $\int (a+b\,\text{Sin}[e+f\,x])^m \left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right) dx$ when $A-B+C=0\ \land\ 2\,m\notin\mathbb{Z}$

Derivation: Algebraic expansion

Basis: If $A-B+C=0$**, then** $A+B\,z+C\,z^2=(A-C)\,(1+z)+C\,(1+z)^2$

Rule: If $A-B+C=0\ \land\ 2\,m\notin\mathbb{Z}$**, then**

$$\int (a+b\,\text{Sin}[e+f\,x])^m \left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right) dx \longrightarrow$$
$$(A-C)\int (a+b\,\text{Sin}[e+f\,x])^m\,(1+\text{Sin}[e+f\,x])\,dx+C\int (a+b\,\text{Sin}[e+f\,x])^m\,(1+\text{Sin}[e+f\,x])^2\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  (A-C)*Int[(a+b*Sin[e+f*x])^m*(1+Sin[e+f*x]),x] + C*Int[(a+b*Sin[e+f*x])^m*(1+Sin[e+f*x])^2,x] /;
FreeQ[{a,b,e,f,A,B,C,m},x] && EqQ[A-B+C,0] && Not[IntegerQ[2*m]]
```

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  (A-C)*Int[(a+b*Sin[e+f*x])^m*(1+Sin[e+f*x]),x] + C*Int[(a+b*Sin[e+f*x])^m*(1+Sin[e+f*x])^2,x] /;
FreeQ[{a,b,e,f,A,C,m},x] && EqQ[A+C,0] && Not[IntegerQ[2*m]]
```

5. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m \left(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2\right) dx$ when $m<-1$

1: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m \left(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2\right) dx$ when $m<-1 \land a^2-b^2==0$

Derivation: Symmetric sine recurrence 2a with $m \to 0$ plus rule for integrands of the form $\mathrm{Sin}[e+f\,x]^2\,(a+b\,\mathrm{Sin}[e+f\,x])^m$

Rule: If $m<-1 \land a^2-b^2==0$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m \left(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2\right) dx \longrightarrow$$

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx+C\int \mathrm{Sin}[e+f\,x]^2\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx \longrightarrow$$

$$\frac{(A\,b-a\,B+b\,C)\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^m}{a\,f\,(2\,m+1)}+$$

$$\frac{1}{a^2\,(2\,m+1)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(a\,A\,(m+1)+m\,(b\,B-a\,C)+b\,C\,(2\,m+1)\,\mathrm{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  (A*b-a*B+b*C)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m/(a*f*(2*m+1)) +
  1/(a^2*(2*m+1))*Int[(a+b*Sin[e+f*x])^(m+1)*Simp[a*A*(m+1)+m*(b*B-a*C)+b*C*(2*m+1)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C},x] && LtQ[m,-1] && EqQ[a^2-b^2,0]
```

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  b*(A+C)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m/(a*f*(2*m+1)) +
  1/(a^2*(2*m+1))*Int[(a+b*Sin[e+f*x])^(m+1)*Simp[a*A*(m+1)-a*C*m+b*C*(2*m+1)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C},x] && LtQ[m,-1] && EqQ[a^2-b^2,0]
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m \left(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2\right) dx$ when $m<-1 \land a^2-b^2\neq 0$

Derivation: Nondegenerate sine recurrence 1a with $n \to 0,\ p \to 0$

Rule: If $m<-1 \land a^2-b^2\neq 0$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m \left(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2\right) dx \longrightarrow$$

$$-\frac{\left(A\,b^2-a\,b\,B+a^2\,C\right)\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}}{b\,f\,(m+1)\,\left(a^2-b^2\right)}+$$

$$\frac{1}{b\,(m+1)\,(a^2-b^2)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\left(b\,(a\,A-b\,B+a\,C)\,(m+1)-\left(A\,b^2-a\,b\,B+a^2\,C+b\,(A\,b-a\,B+b\,C)\,(m+1)\right)\mathrm{Sin}[e+f\,x]\right)dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -(A*b^2-a*b*B+a^2*C)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b*f*(m+1)*(a^2-b^2)) +
  1/(b*(m+1)*(a^2-b^2))*
    Int[(a+b*Sin[e+f*x])^(m+1)*Simp[b*(a*A-b*B+a*C)*(m+1)-(A*b^2-a*b*B+a^2*C+b*(A*b-a*B+b*C)*(m+1))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C},x] && LtQ[m,-1] && NeQ[a^2-b^2,0]
```

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -(A*b^2+a^2*C)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b*f*(m+1)*(a^2-b^2)) +
  1/(b*(m+1)*(a^2-b^2))*
    Int[(a+b*Sin[e+f*x])^(m+1)*Simp[a*b*(A+C)*(m+1)-(A*b^2+a^2*C+b^2*(A+C)*(m+1))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C},x] && LtQ[m,-1] && NeQ[a^2-b^2,0]
```

6: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\left(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2\right)dx$ when $m \nless -1$

Derivation: Nondegenerate sine recurrence 1b with $m\to 0,\ p\to 0$

Rule: If $m \nless -1$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\left(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2\right)dx\longrightarrow$$
$$-\frac{C\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}}{b\,f\,(m+2)}+\frac{1}{b\,(m+2)}\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(A\,b\,(m+2)+b\,C\,(m+1)+(b\,B\,(m+2)-a\,C)\,\mathrm{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_.*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b*f*(m+2)) +
  1/(b*(m+2))*Int[(a+b*Sin[e+f*x])^m*Simp[A*b*(m+2)+b*C*(m+1)+(b*B*(m+2)-a*C)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C,m},x] && Not[LtQ[m,-1]]
```

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b*f*(m+2)) +
  1/(b*(m+2))*Int[(a+b*Sin[e+f*x])^m*Simp[A*b*(m+2)+b*C*(m+1)-a*C*Sin[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C,m},x] && Not[LtQ[m,-1]]
```

Rules for integrands of the form $(b \sin[e+f x]^p)^m \left(A + B \sin[e+f x] + C \sin[e+f x]^2\right)$

1: $\int (b \sin[e+f x]^p)^m \left(A + B \sin[e+f x] + C \sin[e+f x]^2\right) dx$ when $m \notin \mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \frac{(b \sin[e+f x]^p)^m}{(b \sin[e+f x])^{m p}} == 0$
- **Rule: If** $m \notin \mathbb{Z}$, **then**

$$\int (b \sin[e+f x]^p)^m \left(A + B \sin[e+f x] + C \sin[e+f x]^2\right) dx \longrightarrow \frac{(b \sin[e+f x]^p)^m}{(b \sin[e+f x])^{m p}} \int (b \sin[e+f x])^{m p} \left(A + B \sin[e+f x] + C \sin[e+f x]^2\right) dx$$

- **Program code:**

```
Int[(b_.*sin[e_.+f_.*x_]^p_)^m_*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  (b*Sin[e+f*x]^p)^m/(b*Sin[e+f*x])^(m*p)*Int[(b*Sin[e+f*x])^(m*p)*(A+B*Sin[e+f*x]+C*Sin[e+f*x]^2),x] /;
FreeQ[{b,e,f,A,B,C,m,p},x] && Not[IntegerQ[m]]
```

```
Int[(b_.*cos[e_.+f_.*x_]^p_)^m_*(A_.+B_.*cos[e_.+f_.*x_]+C_.*cos[e_.+f_.*x_]^2),x_Symbol] :=
  (b*Cos[e+f*x]^p)^m/(b*Cos[e+f*x])^(m*p)*Int[(b*Cos[e+f*x])^(m*p)*(A+B*Cos[e+f*x]+C*Cos[e+f*x]^2),x] /;
FreeQ[{b,e,f,A,B,C,m,p},x] && Not[IntegerQ[m]]
```

```
Int[(b_.*sin[e_.+f_.*x_]^p_)^m_*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  (b*Sin[e+f*x]^p)^m/(b*Sin[e+f*x])^(m*p)*Int[(b*Sin[e+f*x])^(m*p)*(A+C*Sin[e+f*x]^2),x] /;
FreeQ[{b,e,f,A,C,m,p},x] && Not[IntegerQ[m]]
```

```
Int[(b_.*cos[e_.+f_.*x_]^p_)^m_*(A_.+C_.*cos[e_.+f_.*x_]^2),x_Symbol] :=
  (b*Cos[e+f*x]^p)^m/(b*Cos[e+f*x])^(m*p)*Int[(b*Cos[e+f*x])^(m*p)*(A+C*Cos[e+f*x]^2),x] /;
FreeQ[{b,e,f,A,C,m,p},x] && Not[IntegerQ[m]]
```