Rules for integrands of the form $u\left(a+b\,\mathrm{Sec}[e+f\,x]^2\right)^p$ when $a+b==0$

1: $\int u\left(a+b\,\mathrm{Sec}[e+f\,x]^2\right)^p\,dx$ when $a+b==0\,\land\,p\in\mathbb{Z}$

- **Derivation: Algebraic simplification**
- **Basis: If $a+b==0$, then $a+b\,\mathrm{Sec}[z]^2==b\,\mathrm{Tan}[z]^2$**
- **Rule: If $a+b==0\,\land\,p\in\mathbb{Z}$, then**

$$\int u\left(a+b\,\mathrm{Sec}[e+f\,x]^2\right)^p\,dx\;\longrightarrow\;b^p\int u\,\mathrm{Tan}[e+f\,x]^{2\,p}\,dx$$

- **Program code:**

```
Int[u_.*(a_+b_.*sec[e_.+f_.*x_]^2)^p_,x_Symbol] :=
  b^p*Int[ActivateTrig[u*tan[e+f*x]^(2*p)],x] /;
FreeQ[{a,b,e,f,p},x] && EqQ[a+b,0] && IntegerQ[p]
```

2: $\int u\left(a+b\,\mathrm{Sec}[e+f\,x]^2\right)^p\,dx$ when $a+b==0$

- **Derivation: Algebraic simplification**
- **Basis: If $a+b==0$, then $a+b\,\mathrm{Sec}[z]^2==b\,\mathrm{Tan}[z]^2$**
- **Rule: If $a+b==0$, then**

$$\int u\left(a+b\,\mathrm{Sec}[e+f\,x]^2\right)^p\,dx\;\longrightarrow\;\int u\left(b\,\mathrm{Tan}[e+f\,x]^2\right)^p\,dx$$

- **Program code:**

```
Int[u_.*(a_+b_.*sec[e_.+f_.*x_]^2)^p_,x_Symbol] :=
  Int[ActivateTrig[u*(b*tan[e+f*x]^2)^p],x] /;
FreeQ[{a,b,e,f,p},x] && EqQ[a+b,0]
```

Rules for integrands of the form $(d\,\text{Trig}[e+f\,x])^m\,(a+b\,(c\,\text{Sec}[e+f\,x])^n)^p$

1. $\int (d\,\text{Trig}[e+f\,x])^m\,(b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx$ when $p\notin\mathbb{Z}$

   1. $\int (b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx$ when $p\notin\mathbb{Z}$

      1: $\int \left(b\,\text{Sec}[e+f\,x]^2\right)^p\,dx$ when $p\notin\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Sec}[z]^2 == 1+\text{Tan}[z]^2$

Basis: $F\left[\text{Sec}[e+f\,x]^2\right] == \frac{1}{f}\,\text{Subst}\left[\frac{F[1+x^2]}{1+x^2},\,x,\,\text{Tan}[e+f\,x]\right]\,\partial_x\text{Tan}[e+f\,x]$

Rule: If $p\notin\mathbb{Z}$**, then**

$$\int \left(b\,\text{Sec}[e+f\,x]^2\right)^p\,dx \longrightarrow \frac{b}{f}\,\text{Subst}\left[\int \left(b+b\,x^2\right)^{p-1}\,dx,\,x,\,\text{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(b_.*sec[e_.+f_.*x_]^2)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  b*ff/f*Subst[Int[(b+b*ff^2*x^2)^(p-1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{b,e,f,p},x] && Not[IntegerQ[p]]
```

2: $\int (b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx$ when $p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(b\,F[x]^n)^p}{F[x]^{n\,p}} == 0$

- **Rule: If** $p \notin \mathbb{Z}$**, then**

$$\int (b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx \longrightarrow \frac{b^{\mathrm{IntPart}[p]}\,(b\,(c\,\mathrm{Sec}[e+f\,x])^n)^{\mathrm{FracPart}[p]}}{(c\,\mathrm{Sec}[e+f\,x])^{n\,\mathrm{FracPart}[p]}} \int (c\,\mathrm{Sec}[e+f\,x])^{n\,p}\,dx$$

- **Program code:**

```
Int[(b_.*(c_.*sec[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  b^IntPart[p]*(b*(c*Sec[e+f*x])^n)^FracPart[p]/(c*Sec[e+f*x])^(n*FracPart[p])*Int[(c*Sec[e+f*x])^(n*p),x] /;
FreeQ[{b,c,e,f,n,p},x] && Not[IntegerQ[p]]
```

2. $\int (b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx$ when $p \notin \mathbb{Z}$

1: $\int \mathrm{Tan}[e+f\,x]^m\,\left(b\,\mathrm{Sec}[e+f\,x]^2\right)^p\,dx$ when $p \notin \mathbb{Z}\ \bigwedge\ \frac{m-1}{2} \in \mathbb{Z}$

- **Derivation: Integration by substitution**

- **Basis:** If $\frac{m-1}{2} \in \mathbb{Z}$, then $\mathrm{Tan}[e+f\,x]^m\,F\left[\mathrm{Sec}[e+f\,x]^2\right] == \frac{1}{2\,f}\,\mathrm{Subst}\left[\frac{(-1+x)^{\frac{m-1}{2}}\,F[x]}{x},\ x,\ \mathrm{Sec}[e+f\,x]^2\right]\,\partial_x \mathrm{Sec}[e+f\,x]^2$

- **Rule: If** $p \notin \mathbb{Z}\ \bigwedge\ \frac{m-1}{2} \in \mathbb{Z}$**, then**

$$\int \mathrm{Tan}[e+f\,x]^m\,\left(b\,\mathrm{Sec}[e+f\,x]^2\right)^p\,dx \longrightarrow \frac{b}{2\,f}\,\mathrm{Subst}\left[\int (-1+x)^{\frac{m-1}{2}}\,(b\,x)^{p-1}\,dx,\ x,\ \mathrm{Sec}[e+f\,x]^2\right]$$

- **Program code:**

```
Int[tan[e_.+f_.*x_]^m_.*(b_.*sec[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  b/(2*f)*Subst[Int[(-1+x)^((m-1)/2)*(b*x)^(p-1),x],x,Sec[e+f*x]^2] /;
FreeQ[{b,e,f,p},x] && Not[IntegerQ[p]] && IntegerQ[(m-1)/2]
```

2: $\int u\,(b\,\text{Sec}[e+f\,x]^n)^p\,dx$ when $p \notin \mathbb{Z} \wedge n \in \mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(b\,\text{Sec}[e+f\,x]^n)^p}{\text{Sec}[e+f\,x]^{n\,p}} == 0$

Rule: If $p \notin \mathbb{Z} \wedge n \in \mathbb{Z}$, **then**

$$\int u\,(b\,\text{Sec}[e+f\,x]^n)^p\,dx \longrightarrow \frac{b^{\text{IntPart}[p]}\,(b\,\text{Sec}[e+f\,x]^n)^{\text{FracPart}[p]}}{\text{Sec}[e+f\,x]^{n\,\text{FracPart}[p]}}\int u\,\text{Sec}[e+f\,x]^{n\,p}\,dx$$

Program code:

```
Int[u_.*(b_.*sec[e_.+f_.*x_]^n_)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Sec[e+f*x],x]},
  (b*ff^n)^IntPart[p]*(b*Sec[e+f*x]^n)^FracPart[p]/(Sec[e+f*x]/ff)^(n*FracPart[p])*
    Int[ActivateTrig[u]*(Sec[e+f*x]/ff)^(n*p),x]] /;
FreeQ[{b,e,f,n,p},x] && Not[IntegerQ[p]] && IntegerQ[n] &&
  (EqQ[u,1] || MatchQ[u,(d_.*trig_[e+f*x])^m_. /; FreeQ[{d,m},x] && MemberQ[{sin,cos,tan,cot,sec,csc},trig]])
```

3: $\int u\,(b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx$ when $p \notin \mathbb{Z} \wedge n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(b\,(c\,\text{Sec}[e+f\,x])^n)^p}{(c\,\text{Sec}[e+f\,x])^{n\,p}} == 0$

Rule: If $p \notin \mathbb{Z} \wedge n \notin \mathbb{Z}$, **then**

$$\int (b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx \longrightarrow \frac{b^{\text{IntPart}[p]}\,(b\,(c\,\text{Sec}[e+f\,x])^n)^{\text{FracPart}[p]}}{(c\,\text{Sec}[e+f\,x])^{n\,\text{FracPart}[p]}}\int (c\,\text{Sec}[e+f\,x])^{n\,p}\,dx$$

Program code:

```
Int[u_.*(b_.*(c_.*sec[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  b^IntPart[p]*(b*(c*Sec[e+f*x])^n)^FracPart[p]/(c*Sec[e+f*x])^(n*FracPart[p])*
    Int[ActivateTrig[u]*(c*Sec[e+f*x])^(n*p),x] /;
FreeQ[{b,c,e,f,n,p},x] && Not[IntegerQ[p]] && Not[IntegerQ[n]] &&
  (EqQ[u,1] || MatchQ[u,(d_.*trig_[e+f*x])^m_. /; FreeQ[{d,m},x] && MemberQ[{sin,cos,tan,cot,sec,csc},trig]])
```

2. $\int (a+b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx$

1. $\int \left(a+b\,\mathrm{Sec}[e+f\,x]^2\right)^p\,dx$

1: $\int \frac{1}{a+b\,\mathrm{Sec}[e+f\,x]^2}\,dx$ when $a+b\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+b\,\mathrm{Sec}[z]^2} == \frac{1}{a}-\frac{b}{a\,\left(b+a\,\mathrm{Cos}[z]^2\right)}$

Rule: If $a+b\neq 0$, then

$$\int \frac{1}{a+b\,\mathrm{Sec}[e+f\,x]^2}\,dx \longrightarrow \frac{x}{a}-\frac{b}{a}\int \frac{1}{b+a\,\mathrm{Cos}[e+f\,x]^2}\,dx$$

Program code:

```
Int[1/(a_+b_.*sec[e_.+f_.*x_]^2),x_Symbol] :=
  x/a - b/a*Int[1/(b+a*Cos[e+f*x]^2),x] /;
FreeQ[{a,b,e,f},x] && NeQ[a+b,0]
```

2: $\int \left(a+b\,\mathrm{Sec}[e+f\,x]^2\right)^p\,dx$ when $a+b\neq 0\,\wedge\,p\neq -1$

Derivation: Integration by substitution

Basis: $F\left[\mathrm{Sec}[e+f\,x]^2\right] == \frac{1}{f}\,\mathrm{Subst}\left[\frac{F[1+x^2]}{1+x^2},\,x,\,\mathrm{Tan}[e+f\,x]\right]\,\partial_x\mathrm{Tan}[e+f\,x]$

Rule: If $a+b\neq 0\,\wedge\,p\neq -1$, then

$$\int \left(a+b\,\mathrm{Sec}[e+f\,x]^2\right)^p\,dx \longrightarrow \frac{1}{f}\,\mathrm{Subst}\left[\int \frac{\left(a+b+b\,x^2\right)^p}{1+x^2}\,dx,\,x,\,\mathrm{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(a_+b_.*sec[e_.+f_.*x_]^2)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[(a+b+b*ff^2*x^2)^p/(1+ff^2*x^2),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && NeQ[a+b,0] && NeQ[p,-1]
```

2: $\int (a + b \operatorname{Sec}[e + f x]^4)^p \, dx$ when $2 p \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $F[\operatorname{Sec}[e + f x]^2] = \frac{1}{f} \operatorname{Subst}\left[\frac{F[1+x^2]}{1+x^2}, x, \operatorname{Tan}[e + f x]\right] \partial_x \operatorname{Tan}[e + f x]$

Rule: If $2 p \in \mathbb{Z}$, then

$$\int (a + b \operatorname{Sec}[e + f x]^4)^p \, dx \longrightarrow \frac{1}{f} \operatorname{Subst}\left[\int \frac{(a + b + 2 b x^2 + b x^4)^p}{1 + x^2} \, dx, x, \operatorname{Tan}[e + f x]\right]$$

Program code:

```
Int[(a_+b_.*sec[e_.+f_.*x_]^4)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[(a+b+2*b*ff^2*x^2+b*ff^4*x^4)^p/(1+ff^2*x^2),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[2*p]
```

3: $\int (a + b \operatorname{Sec}[e + f x]^n)^p \, dx$ when $\frac{n}{2} \in \mathbb{Z} \bigwedge p + 2 \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F[\operatorname{Sec}[e + f x]^2] = \frac{1}{f} \operatorname{Subst}\left[\frac{F[1+x^2]}{1+x^2}, x, \operatorname{Tan}[e + f x]\right] \partial_x \operatorname{Tan}[e + f x]$

Rule: If $\frac{n}{2} \in \mathbb{Z} \bigwedge p + 2 \in \mathbb{Z}^+$, then

$$\int (a + b \operatorname{Sec}[e + f x]^n)^p \, dx \longrightarrow \frac{1}{f} \operatorname{Subst}\left[\int \frac{\left(a + b \left(1 + x^2\right)^{n/2}\right)^p}{1 + x^2} \, dx, x, \operatorname{Tan}[e + f x]\right]$$

Program code:

```
Int[(a_+b_.*sec[e_.+f_.*x_]^n_)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[(a+b*(1+ff^2*x^2)^(n/2))^p/(1+ff^2*x^2),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[n/2] && IGtQ[p,-2]
```

X: $\int (a + b\,(c\,\text{Sec}[e + f\,x])^n)^p\,dx$

Rule:

$$\int (a + b\,(c\,\text{Sec}[e + f\,x])^n)^p\,dx \longrightarrow \int (a + b\,(c\,\text{Sec}[e + f\,x])^n)^p\,dx$$

Program code:

```
Int[(a_+b_.*(c_.*sec[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  Unintegrable[(a+b*(c*Sec[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,e,f,n,p},x]
```

3. $\int (d\,\text{Sin}[e + f\,x])^m\,(a + b\,(c\,\text{Sec}[e + f\,x])^n)^p\,dx$

1: $\int \text{Sin}[e + f\,x]^m\,(a + b\,\text{Sec}[e + f\,x]^n)^p\,dx$ when $\frac{m}{2} \in \mathbb{Z} \bigwedge \frac{n}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

- Basis: $\text{Sin}[z]^2 = \frac{\text{Tan}[z]^2}{1+\text{Tan}[z]^2}$
- Basis: $\text{Sec}[z]^2 = 1 + \text{Tan}[z]^2$
- Basis: If $\frac{m}{2} \in \mathbb{Z}$, then $\text{Sin}[e + f\,x]^m\,F\left[\text{Sec}[e + f\,x]^2\right] = \frac{1}{f}\,\text{Subst}\left[\frac{x^m\,F[1+x^2]}{(1+x^2)^{m/2+1}},\ x,\ \text{Tan}[e + f\,x]\right]\,\partial_x \text{Tan}[e + f\,x]$
- Rule: If $\frac{m}{2} \in \mathbb{Z} \bigwedge \frac{n}{2} \in \mathbb{Z}$, then

$$\int \text{Sin}[e + f\,x]^m\,(a + b\,\text{Sec}[e + f\,x]^n)^p\,dx \longrightarrow \frac{1}{f}\,\text{Subst}\left[\int \frac{x^m\,\left(a + b\,\left(1 + x^2\right)^{n/2}\right)^p}{\left(1 + x^2\right)^{m/2+1}}\,dx,\ x,\ \text{Tan}[e + f\,x]\right]$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_*(a_+b_.*sec[e_.+f_.*x_]^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff^(m+1)/f*Subst[Int[x^m*ExpandToSum[a+b*(1+ff^2*x^2)^(n/2),x]^p/(1+ff^2*x^2)^(m/2+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[m/2] && IntegerQ[n/2]
```

2. $\int \text{Sin}[e+f\,x]^m\,(a+b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}$

1: $\int \text{Sin}[e+f\,x]^m\,(a+b\,\text{Sec}[e+f\,x]^n)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\bigwedge n\in\mathbb{Z}\bigwedge p\in\mathbb{Z}$

Derivation: Integration by substitution

- **Basis: If** $\frac{m-1}{2}\in\mathbb{Z}$**, then** $\text{Sin}[e+f\,x]^m\,F[\text{Sec}[e+f\,x]]==-\frac{1}{f}\,\text{Subst}\Big[\left(1-x^2\right)^{\frac{m-1}{2}}\,F\big[\frac{1}{x}\big],\,x,\,\text{Cos}[e+f\,x]\Big]\,\partial_x\text{Cos}[e+f\,x]$

- **Rule: If** $\frac{m-1}{2}\in\mathbb{Z}\bigwedge n\in\mathbb{Z}\bigwedge p\in\mathbb{Z}$**, then**

$$\int \text{Sin}[e+f\,x]^m\,(a+b\,\text{Sec}[e+f\,x]^n)^p\,dx\longrightarrow-\frac{1}{f}\,\text{Subst}\Big[\int\frac{\left(1-x^2\right)^{\frac{m-1}{2}}\,(b+a\,x^n)^p}{x^{n\,p}}\,dx,\,x,\,\text{Cos}[e+f\,x]\Big]$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_.*(a_+b_.*sec[e_.+f_.*x_]^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Cos[e+f*x],x]},
  -ff/f*Subst[Int[(1-ff^2*x^2)^((m-1)/2)*(b+a*(ff*x)^n)^p/(ff*x)^(n*p),x],x,Cos[e+f*x]/ff]] /;
FreeQ[{a,b,e,f},x] && IntegerQ[(m-1)/2] && IntegerQ[n] && IntegerQ[p]
```

2: $\int \text{Sin}[e+f\,x]^m\,(a+b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\bigwedge(m>0\bigvee n==2\bigvee n==4)$

Derivation: Integration by substitution

- **Basis: If** $\frac{m-1}{2}\in\mathbb{Z}$**, then** $\text{Sin}[e+f\,x]^m\,F[\text{Sec}[e+f\,x]]==\frac{1}{f}\,\text{Subst}\Big[\frac{(-1+x^2)^{\frac{m-1}{2}}\,F[x]}{x^{m+1}},\,x,\,\text{Sec}[e+f\,x]\Big]\,\partial_x\text{Sec}[e+f\,x]$

- **Rule: If** $\frac{m-1}{2}\in\mathbb{Z}\bigwedge(m>0\lor n==2\lor n==4)$**, then**

$$\int \text{Sin}[e+f\,x]^m\,(a+b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx\longrightarrow\frac{1}{f}\,\text{Subst}\Big[\int\frac{\left(-1+x^2\right)^{\frac{m-1}{2}}\,(a+b\,(c\,x)^n)^p}{x^{m+1}}\,dx,\,x,\,\text{Sec}[e+f\,x]\Big]$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_.*(a_+b_.*(c_.*sec[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Cos[e+f*x],x]},
  1/(f*ff^m)*Subst[Int[(-1+ff^2*x^2)^((m-1)/2)*(a+b*(c*ff*x)^n)^p/x^(m+1),x],x,Sec[e+f*x]/ff]] /;
FreeQ[{a,b,c,e,f,n,p},x] && IntegerQ[(m-1)/2] && (GtQ[m,0] || EqQ[n,2] || EqQ[n,4])
```

X: $\int (d\,\text{Sin}[e+f\,x])^m\,(a+b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx$

Rule:

$$\int (d\,\text{Sin}[e+f\,x])^m\,(a+b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx \longrightarrow \int (d\,\text{Sin}[e+f\,x])^m\,(a+b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx$$

Program code:

```
Int[(d_.*sin[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*sec[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Unintegrable[(d*Sin[e+f*x])^m*(a+b*(c*Sec[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x]
```

4. $\int (d\,\text{Cos}[e+f\,x])^m\,(a+b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx$

1: $\int (d\,\text{Cos}[e+f\,x])^m\,(a+b\,\text{Sec}[e+f\,x]^n)^p\,dx$ when $m\notin\mathbb{Z}\,\land\,(n\mid p)\in\mathbb{Z}$

Derivation: Algebraic normalization

Basis: If $(n\mid p)\in\mathbb{Z}$**, then** $(a+b\,\text{Sec}[e+f\,x]^n)^p == d^{n\,p}\,(d\,\text{Cos}[e+f\,x])^{-n\,p}\,(b+a\,\text{Cos}[e+f\,x]^n)^p$

Rule: If $m\notin\mathbb{Z}\,\land\,(n\mid p)\in\mathbb{Z}$**, then**

$$\int (d\,\text{Cos}[e+f\,x])^m\,(a+b\,\text{Sec}[e+f\,x]^n)^p\,dx \longrightarrow d^{n\,p}\int (d\,\text{Cos}[e+f\,x])^{m-n\,p}\,(b+a\,\text{Cos}[e+f\,x]^n)^p\,dx$$

Program code:

```
Int[(d_.*cos[e_.+f_.*x_])^m_*(a_+b_.*sec[e_.+f_.*x_]^n_.)^p_.,x_Symbol] :=
  d^(n*p)*Int[(d*Cos[e+f*x])^(m-n*p)*(b+a*Cos[e+f*x]^n)^p,x] /;
FreeQ[{a,b,d,e,f,m,n,p},x] && Not[IntegerQ[m]] && IntegersQ[n,p]
```

2: $\int (d\,\text{Cos}[e+f\,x])^m\,(a+b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx$ when $m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((d\,\text{Cos}[e+f\,x])^m\left(\frac{\text{Sec}[e+f\,x]}{d}\right)^m\right) == 0$

Rule: If $m\notin\mathbb{Z}$**, then**

$$\int (d\,\text{Cos}[e+f\,x])^m\,(a+b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx \longrightarrow (d\,\text{Cos}[e+f\,x])^{\text{FracPart}[m]}\left(\frac{\text{Sec}[e+f\,x]}{d}\right)^{\text{FracPart}[m]}\int\left(\frac{\text{Sec}[e+f\,x]}{d}\right)^{-m}(a+b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx$$

Program code:

```
Int[(d_.*cos[e_.+f_.*x_])^m_*(a_+b_.*(c_.*sec[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  (d*Cos[e+f*x])^FracPart[m]*(Sec[e+f*x]/d)^FracPart[m]*Int[(Sec[e+f*x]/d)^(-m)*(a+b*(c*Sec[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]]
```

5. $\int (d\,\text{Tan}[e+f\,x])^m\,(a+b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx$

1. $\int \text{Tan}[e+f\,x]^m\,(a+b\,(c\,\text{Sec}[e+f\,x])^n)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}$

1: $\int \text{Tan}[e+f\,x]^m\,(a+b\,\text{Sec}[e+f\,x]^n)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\bigwedge n\in\mathbb{Z}\bigwedge p\in\mathbb{Z}$

Derivation: Integration by substitution

- **Basis:** $\text{Tan}[z]^2 = \frac{1-\text{Cos}[z]^2}{\text{Cos}[z]^2}$
- **Basis: If** $\frac{m-1}{2}\in\mathbb{Z}$**, then** $\text{Tan}[e+f\,x]^m\,F[\text{Sec}[e+f\,x]] = -\frac{1}{f}\,\text{Subst}\left[\frac{(1-x^2)^{\frac{m-1}{2}}\,F\left[\frac{1}{x}\right]}{x^m},\,x,\,\text{Cos}[e+f\,x]\right]\partial_x\text{Cos}[e+f\,x]$
- **Rule: If** $\frac{m-1}{2}\in\mathbb{Z}\bigwedge n\in\mathbb{Z}\bigwedge p\in\mathbb{Z}$**, then**

$$\int \text{Tan}[e+f\,x]^m\,(a+b\,\text{Sec}[e+f\,x]^n)^p\,dx \longrightarrow -\frac{1}{f}\,\text{Subst}\left[\int\frac{\left(1-x^2\right)^{\frac{m-1}{2}}(b+a\,x^n)^p}{x^{m+n\,p}}\,dx,\,x,\,\text{Cos}[e+f\,x]\right]$$

Program code:

```
Int[tan[e_.+f_.*x_]^m_.*(a_+b_.*sec[e_.+f_.*x_]^n_)^p_.,x_Symbol] :=
  Module[{ff=FreeFactors[Cos[e+f*x],x]},
  -1/(f*ff^(m+n*p-1))*Subst[Int[(1-ff^2*x^2)^((m-1)/2)*(b+a*(ff*x)^n)^p/x^(m+n*p),x],x,Cos[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,n},x] && IntegerQ[(m-1)/2] && IntegerQ[n] && IntegerQ[p]
```

2: $\int \tan[e+f x]^m \left(a+b\,(c\,\sec[e+f x])^n\right)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\bigwedge\left(m>0\bigvee n==2\bigvee n==4\bigvee p\in\mathbb{Z}^+\bigvee(2\,n\mid p)\in\mathbb{Z}\right)$

Derivation: Integration by substitution

- **Basis:** $\tan[z]^2==-1+\sec[z]^2$

- **Basis: If** $\frac{m-1}{2}\in\mathbb{Z}$, **then** $\tan[e+f x]^m\,F[\sec[e+f x]]==\frac{1}{f}\,\text{Subst}\left[\frac{(-1+x^2)^{\frac{m-1}{2}}\,F[x]}{x},\,x,\,\sec[e+f x]\right]\partial_x\sec[e+f x]$

- **Rule: If** $\frac{m-1}{2}\in\mathbb{Z}\bigwedge\left(m>0\bigvee n==2\bigvee n==4\bigvee p\in\mathbb{Z}^+\bigvee(2\,n\mid p)\in\mathbb{Z}\right)$, **then**

$$\int \tan[e+f x]^m \left(a+b\,(c\,\sec[e+f x])^n\right)^p\,dx \rightarrow \frac{1}{f}\,\text{Subst}\left[\int\frac{\left(-1+x^2\right)^{\frac{m-1}{2}}\left(a+b\,(c\,x)^n\right)^p}{x}\,dx,\,x,\,\sec[e+f x]\right]$$

- **Program code:**

```
Int[tan[e_.+f_.*x_]^m_.*(a_+b_.*(c_.*sec[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sec[e+f*x],x]},
  1/f*Subst[Int[(-1+ff^2*x^2)^((m-1)/2)*(a+b*(c*ff*x)^n)^p/x,x],x,Sec[e+f*x]/ff]] /;
FreeQ[{a,b,c,e,f,n,p},x] && IntegerQ[(m-1)/2] && (GtQ[m,0] || EqQ[n,2] || EqQ[n,4] || IGtQ[p,0] || IntegersQ[2*n,p])
```

2. $\int (d\,\mathrm{Tan}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx$

1: $\int (d\,\mathrm{Tan}[e+f\,x])^m\,\left(b\,\mathrm{Sec}[e+f\,x]^2\right)^p\,dx$

Derivation: Integration by substitution

Basis: $\mathrm{Sec}[z]^2 == 1+\mathrm{Tan}[z]^2$

Basis: $(d\,\mathrm{Tan}[e+f\,x])^m\,F\left[\mathrm{Sec}[e+f\,x]^2\right] == \frac{1}{f}\,\mathrm{Subst}\left[\frac{(d\,x)^m\,F[1+x^2]}{1+x^2},\,x,\,\mathrm{Tan}[e+f\,x]\right]\,\partial_x \mathrm{Tan}[e+f\,x]$

Rule:

$$\int (d\,\mathrm{Tan}[e+f\,x])^m\,\left(b\,\mathrm{Sec}[e+f\,x]^2\right)^p\,dx \longrightarrow \frac{b}{f}\,\mathrm{Subst}\left[\int (d\,x)^m\,\left(b+b\,x^2\right)^{p-1}\,dx,\,x,\,\mathrm{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(d_.*tan[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  b*ff/f*Subst[Int[(d*ff*x)^m*(b+b*ff^2*x^2)^(p-1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{b,d,e,f,m,p},x]
```

2: $\int (d\,\mathrm{Tan}[e+f\,x])^m\,(a+b\,\mathrm{Sec}[e+f\,x]^n)^p\,dx$ when $\frac{n}{2}\in\mathbb{Z}\bigwedge\left(\frac{m}{2}\in\mathbb{Z}\bigvee n==2\right)$

Derivation: Integration by substitution

Basis: $\mathrm{Sec}[z]^2 == 1+\mathrm{Tan}[z]^2$

Basis: $(d\,\mathrm{Tan}[e+f\,x])^m\,F\left[\mathrm{Sec}[e+f\,x]^2\right] == \frac{1}{f}\,\mathrm{Subst}\left[\frac{(d\,x)^m\,F[1+x^2]}{1+x^2},\,x,\,\mathrm{Tan}[e+f\,x]\right]\,\partial_x \mathrm{Tan}[e+f\,x]$

Rule: If $\frac{n}{2}\in\mathbb{Z}\bigwedge\left(\frac{m}{2}\in\mathbb{Z}\bigvee n==2\right)$**, then**

$$\int (d\,\mathrm{Tan}[e+f\,x])^m\,(a+b\,\mathrm{Sec}[e+f\,x]^n)^p\,dx \longrightarrow \frac{1}{f}\,\mathrm{Subst}\left[\int \frac{(d\,x)^m\,\left(a+b\,\left(1+x^2\right)^{n/2}\right)^p}{1+x^2}\,dx,\,x,\,\mathrm{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(d_.*tan[e_.+f_.*x_])^m_*(a_+b_.*sec[e_.+f_.*x_]^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[(d*ff*x)^m*(a+b*(1+ff^2*x^2)^(n/2))^p/(1+ff^2*x^2),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,d,e,f,m,p},x] && IntegerQ[n/2] && (IntegerQ[m/2] || EqQ[n,2])
```

3. $\int (d \operatorname{Tan}[e+f x])^m (b (c \operatorname{Sec}[e+f x])^n)^p \, dx$

1: $\int (d \operatorname{Tan}[e+f x])^m (b (c \operatorname{Sec}[e+f x])^n)^p \, dx$ when $m > 1 \land p n + m - 1 \neq 0$

Reference: G&R 2.510.1

Reference: G&R 2.510.4

Rule: If $m > 1 \land p n + m - 1 \neq 0$, then

$$\int (d \operatorname{Tan}[e+f x])^m (b (c \operatorname{Sec}[e+f x])^n)^p \, dx \longrightarrow$$

$$\frac{d (d \operatorname{Tan}[e+f x])^{m-1} (b (c \operatorname{Sec}[e+f x])^n)^p}{f (p n + m - 1)} - \frac{d^2 (m-1)}{p n + m - 1} \int (d \operatorname{Tan}[e+f x])^{m-2} (b (c \operatorname{Sec}[e+f x])^n)^p \, dx$$

Program code:

```
Int[(d_.*tan[e_.+f_.*x_])^m_*(b_.*(c_.*sec[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  d*(d*Tan[e+f*x])^(m-1)*(b*(c*Sec[e+f*x])^n)^p/(f*(p*n+m-1)) -
  d^2*(m-1)/(p*n+m-1)*Int[(d*Tan[e+f*x])^(m-2)*(b*(c*Sec[e+f*x])^n)^p,x] /;
FreeQ[{b,c,d,e,f,p,n},x] && GtQ[m,1] && NeQ[p*n+m-1,0] && IntegersQ[2*p*n,2*m]
```

2: $\int (d \operatorname{Tan}[e+f x])^m (b (c \operatorname{Sec}[e+f x])^n)^p \, dx$ when $m < -1 \land p n + m + 1 \neq 0$

Reference: G&R 2.510.4

Reference: G&R 2.510.1

Rule: If $m < -1 \land p n + m + 1 \neq 0$, then

$$\int (d \operatorname{Tan}[e+f x])^m (b (c \operatorname{Sec}[e+f x])^n)^p \, dx \longrightarrow$$

$$\frac{(d \operatorname{Tan}[e+f x])^{m+1} (b (c \operatorname{Sec}[e+f x])^n)^p}{d f (m+1)} - \frac{p n + m + 1}{d^2 (m+1)} \int (d \operatorname{Tan}[e+f x])^{m+2} (b (c \operatorname{Sec}[e+f x])^n)^p \, dx$$

Program code:

```
Int[(d_.*tan[e_.+f_.*x_])^m_*(b_.*(c_.*sec[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  (d*Tan[e+f*x])^(m+1)*(b*(c*Sec[e+f*x])^n)^p/(d*f*(m+1)) -
  (p*n+m+1)/(d^2*(m+1))*Int[(d*Tan[e+f*x])^(m+2)*(b*(c*Sec[e+f*x])^n)^p,x] /;
FreeQ[{b,c,d,e,f,p,n},x] && LtQ[m,-1] && NeQ[p*n+m+1,0] && IntegersQ[2*p*n,2*m]
```

U: $\int (d\,\mathrm{Tan}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx$

Rule:

$$\int (d\,\mathrm{Tan}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx \longrightarrow \int (d\,\mathrm{Tan}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx$$

Program code:

```
Int[(d_.*tan[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*sec[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Unintegrable[(d*Tan[e+f*x])^m*(a+b*(c*Sec[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x]
```

6: $\int (d\,\mathrm{Cot}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx$ when $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((d\,\mathrm{Cot}[e+f\,x])^m\left(\frac{\mathrm{Tan}[e+f\,x]}{d}\right)^m\right) == 0$

Rule: If $m \notin \mathbb{Z}$, **then**

$$\int (d\,\mathrm{Cot}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx \longrightarrow (d\,\mathrm{Cot}[e+f\,x])^{\mathrm{FracPart}[m]}\left(\frac{\mathrm{Tan}[e+f\,x]}{d}\right)^{\mathrm{FracPart}[m]}\int\left(\frac{\mathrm{Tan}[e+f\,x]}{d}\right)^{-m}(a+b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx$$

Program code:

```
Int[(d_.*cot[e_.+f_.*x_])^m_*(a_+b_.*(c_.*sec[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  (d*Cot[e+f*x])^FracPart[m]*(Tan[e+f*x]/d)^FracPart[m]*Int[(Tan[e+f*x]/d)^(-m)*(a+b*(c*Sec[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]]
```

7. $\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx$

1: $\int \mathrm{Sec}[e+f\,x]^m\,(a+b\,\mathrm{Sec}[e+f\,x]^n)^p\,dx$ when $\frac{m}{2}\in\mathbb{Z}\bigwedge\frac{n}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\mathrm{Sec}[z]^2 == 1+\mathrm{Tan}[z]^2$

- **Basis: If** $\frac{m}{2}\in\mathbb{Z}$**, then** $\mathrm{Sec}[e+f\,x]^m\,F\left[\mathrm{Sec}[e+f\,x]^2\right] == \frac{1}{f}\,\mathrm{Subst}\left[\left(1+x^2\right)^{\frac{m}{2}-1}F\left[1+x^2\right],\,x,\,\mathrm{Tan}[e+f\,x]\right]\partial_x\mathrm{Tan}[e+f\,x]$
- **Rule: If** $\frac{m}{2}\in\mathbb{Z}\bigwedge\frac{n}{2}\in\mathbb{Z}$**, then**

$$\int \mathrm{Sec}[e+f\,x]^m\,(a+b\,\mathrm{Sec}[e+f\,x]^n)^p\,dx \longrightarrow \frac{1}{f}\,\mathrm{Subst}\left[\int\left(1+x^2\right)^{\frac{m}{2}-1}\left(a+b\left(1+x^2\right)^{n/2}\right)^p dx,\,x,\,\mathrm{Tan}[e+f\,x]\right]$$

Program code:

```
Int[sec[e_.+f_.*x_]^m_*(a_+b_.*sec[e_.+f_.*x_]^n_)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[(1+ff^2*x^2)^(m/2-1)*ExpandToSum[a+b*(1+ff^2*x^2)^(n/2),x]^p,x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[m/2] && IntegerQ[n/2]
```

2. $\int \text{Sec}[e+f x]^m \,(a+b\,\text{Sec}[e+f x]^n)^p \, dx$ when $\frac{m-1}{2} \in \mathbb{Z} \bigwedge \frac{n}{2} \in \mathbb{Z}$

1: $\int \text{Sec}[e+f x]^m \,(a+b\,\text{Sec}[e+f x]^n)^p \, dx$ when $\frac{m-1}{2} \in \mathbb{Z} \bigwedge \frac{n}{2} \in \mathbb{Z} \bigwedge p \in \mathbb{Z}$

Derivation: Integration by substitution

- **Basis:** $\text{Sec}[z]^2 == \frac{1}{1-\text{Sin}[z]^2}$
- **Basis: If** $\frac{m-1}{2} \in \mathbb{Z}$**, then** $\text{Sec}[e+f x]^m \, F\left[\text{Sec}[e+f x]^2\right] == \frac{1}{f} \, \text{Subst}\left[\frac{F\left[\frac{1}{1-x^2}\right]}{(1-x^2)^{\frac{m+1}{2}}}, x, \text{Sin}[e+f x]\right] \partial_x \text{Sin}[e+f x]$
- **Rule: If** $\frac{m-1}{2} \in \mathbb{Z} \bigwedge \frac{n}{2} \in \mathbb{Z} \bigwedge p \in \mathbb{Z}$**, then**

$$\int \text{Sec}[e+f x]^m \,(a+b\,\text{Sec}[e+f x]^n)^p \, dx \longrightarrow \frac{1}{f} \, \text{Subst}\left[\int \frac{\left(b+a\left(1-x^2\right)^{n/2}\right)^p}{\left(1-x^2\right)^{(m+n\,p+1)/2}} \, dx, x, \text{Sin}[e+f x]\right]$$

Program code:

```
Int[sec[e_.+f_.*x_]^m_.*(a_+b_.*sec[e_.+f_.*x_]^n_)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff/f*Subst[Int[ExpandToSum[b+a*(1-ff^2*x^2)^(n/2),x]^p/(1-ff^2*x^2)^((m+n*p+1)/2),x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,e,f},x] && IntegerQ[(m-1)/2] && IntegerQ[n/2] && IntegerQ[p]
```

2: $\int \text{Sec}[e+f\,x]^m\,(a+b\,\text{Sec}[e+f\,x]^n)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\,\bigwedge\,\frac{n}{2}\in\mathbb{Z}\,\bigwedge\,p\notin\mathbb{Z}$

Derivation: Integration by substitution

- **Basis:** $\text{Sec}[z]^2 == \frac{1}{1-\text{Sin}[z]^2}$

- **Basis: If** $\frac{m-1}{2}\in\mathbb{Z}$**, then** $\text{Sec}[e+f\,x]^m\,F\left[\text{Sec}[e+f\,x]^2\right] == \frac{1}{f}\,\text{Subst}\left[\frac{F\left[\frac{1}{1-x^2}\right]}{(1-x^2)^{\frac{m+1}{2}}},\,x,\,\text{Sin}[e+f\,x]\right]\partial_x\text{Sin}[e+f\,x]$

- **Rule: If** $\frac{m-1}{2}\in\mathbb{Z}\,\bigwedge\,\frac{n}{2}\in\mathbb{Z}\,\bigwedge\,p\notin\mathbb{Z}$**, then**

$$\int \text{Sec}[e+f\,x]^m\,(a+b\,\text{Sec}[e+f\,x]^n)^p\,dx\,\longrightarrow\,\frac{1}{f}\,\text{Subst}\left[\int\frac{\left(a+\frac{b}{(1-x^2)^{n/2}}\right)^p}{\left(1-x^2\right)^{\frac{m+1}{2}}}\,dx,\,x,\,\text{Sin}[e+f\,x]\right]$$

Program code:

```
Int[sec[e_.+f_.*x_]^m_.*(a_+b_.*sec[e_.+f_.*x_]^n_)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff/f*Subst[Int[(a+b/(1-ff^2*x^2)^(n/2))^p/(1-ff^2*x^2)^((m+1)/2),x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[(m-1)/2] && IntegerQ[n/2] && Not[IntegerQ[p]]
```

3: $\int \text{Sec}[e+f\,x]^m\,(a+b\,\text{Sec}[e+f\,x]^n)^p\,dx$ when $(m\,|\,n\,|\,p)\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $(m\,|\,n\,|\,p)\in\mathbb{Z}$**, then**

$$\int \text{Sec}[e+f\,x]^m\,(a+b\,\text{Sec}[e+f\,x]^n)^p\,dx\,\longrightarrow\,\int \text{ExpandTrig}[\text{Sec}[e+f\,x]^m\,(a+b\,\text{Sec}[e+f\,x]^n)^p,\,x]\,dx$$

Program code:

```
Int[sec[e_.+f_.*x_]^m_.*(a_+b_.*sec[e_.+f_.*x_]^n_)^p_,x_Symbol] :=
  Int[ExpandTrig[sec[e+f*x]^m*(a+b*sec[e+f*x]^n)^p,x],x] /;
FreeQ[{a,b,e,f},x] && IntegersQ[m,n,p]
```

U: $\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx$

- **Rule:**

$$\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx \longrightarrow \int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx$$

- **Program code:**

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*sec[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Unintegrable[(d*Sec[e+f*x])^m*(a+b*(c*Sec[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x]
```

8: $\int (d\,\mathrm{Csc}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx$ when $m \notin \mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x\left((d\,\mathrm{Csc}[e+f\,x])^m\left(\frac{\mathrm{Sin}[e+f\,x]}{d}\right)^m\right) == 0$
- **Rule: If** $m \notin \mathbb{Z}$, **then**

$$\int (d\,\mathrm{Csc}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx \longrightarrow (d\,\mathrm{Csc}[e+f\,x])^{\mathrm{FracPart}[m]}\left(\frac{\mathrm{Sin}[e+f\,x]}{d}\right)^{\mathrm{FracPart}[m]}\int\left(\frac{\mathrm{Sin}[e+f\,x]}{d}\right)^{-m}(a+b\,(c\,\mathrm{Sec}[e+f\,x])^n)^p\,dx$$

- **Program code:**

```
Int[(d_.*csc[e_.+f_.*x_])^m_*(a_+b_.*(c_.*sec[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  (d*Csc[e+f*x])^FracPart[m]*(Sin[e+f*x]/d)^FracPart[m]*Int[(Sin[e+f*x]/d)^(-m)*(a+b*(c*Sec[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]]
```