Rules for normalizing to known secant integrands

1. $\int u\,(c\,\text{Trig}[a+b\,x])^m\,(d\,\text{Trig}[a+b\,x])^n\,dx$ when KnownSecantIntegrandQ[u, x]

1: $\int u\,(c\,\text{Sin}[a+b\,x])^m\,(d\,\text{Csc}[a+b\,x])^n\,dx$ when KnownSecantIntegrandQ[u, x]

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x\left((c\,\text{Sin}[a+b\,x])^m\,(d\,\text{Csc}[a+b\,x])^m\right) == 0$
- **Rule: If KnownSecantIntegrandQ[u, x], then**

$$\int u\,(c\,\text{Sin}[a+b\,x])^m\,(d\,\text{Csc}[a+b\,x])^n\,dx \longrightarrow (c\,\text{Sin}[a+b\,x])^m\,(d\,\text{Csc}[a+b\,x])^m\int u\,(d\,\text{Csc}[a+b\,x])^{n-m}\,dx$$

- **Program code:**

```
Int[u_*(c_.*sin[a_.+b_.*x_])^m_.*(d_.*csc[a_.+b_.*x_])^n_.,x_Symbol] :=
  (c*Sin[a+b*x])^m*(d*Csc[a+b*x])^m*Int[ActivateTrig[u]*(d*Csc[a+b*x])^(n-m),x] /;
FreeQ[{a,b,c,d,m,n},x] && KnownSecantIntegrandQ[u,x]
```

2: $\int u\,(c\,\text{Cos}[a+b\,x])^m\,(d\,\text{Sec}[a+b\,x])^n\,dx$ when KnownSecantIntegrandQ[u, x]

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x\left((c\,\text{Cos}[a+b\,x])^m\,(d\,\text{Sec}[a+b\,x])^m\right) == 0$
- **Rule: If KnownSecantIntegrandQ[u, x], then**

$$\int u\,(c\,\text{Cos}[a+b\,x])^m\,(d\,\text{Sec}[a+b\,x])^n\,dx \longrightarrow (c\,\text{Cos}[a+b\,x])^m\,(d\,\text{Sec}[a+b\,x])^m\int u\,(d\,\text{Sec}[a+b\,x])^{n-m}\,dx$$

- **Program code:**

```
Int[u_*(c_.*cos[a_.+b_.*x_])^m_.*(d_.*sec[a_.+b_.*x_])^n_.,x_Symbol] :=
  (c*Cos[a+b*x])^m*(d*Sec[a+b*x])^m*Int[ActivateTrig[u]*(d*Sec[a+b*x])^(n-m),x] /;
FreeQ[{a,b,c,d,m,n},x] && KnownSecantIntegrandQ[u,x]
```

3. $\int u\,(c\,\mathrm{Tan}[a+b\,x])^m\,(d\,\mathrm{Trig}[a+b\,x])^n\,dx$ when KnownSecantIntegrandQ[u, x]

1: $\int u\,(c\,\mathrm{Tan}[a+b\,x])^m\,(d\,\mathrm{Sec}[a+b\,x])^n\,dx$ when KnownSecantIntegrandQ[u, x] $\wedge\ m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,\mathrm{Tan}[a+b\,x])^m\,(d\,\mathrm{Csc}[a+b\,x])^m}{(d\,\mathrm{Sec}[a+b\,x])^m} == 0$

Rule: If KnownSecantIntegrandQ[u, x] $\wedge\ m\notin\mathbb{Z}$**, then**

$$\int u\,(c\,\mathrm{Tan}[a+b\,x])^m\,(d\,\mathrm{Sec}[a+b\,x])^n\,dx \longrightarrow \frac{(c\,\mathrm{Tan}[a+b\,x])^m\,(d\,\mathrm{Csc}[a+b\,x])^m}{(d\,\mathrm{Sec}[a+b\,x])^m}\int \frac{u\,(d\,\mathrm{Sec}[a+b\,x])^{m+n}}{(d\,\mathrm{Csc}[a+b\,x])^m}\,dx$$

Program code:

```
Int[u_*(c_.*tan[a_.+b_.*x_])^m_.*(d_.*sec[a_.+b_.*x_])^n_.,x_Symbol] :=
  (c*Tan[a+b*x])^m*(d*Csc[a+b*x])^m/(d*Sec[a+b*x])^m*Int[ActivateTrig[u]*(d*Sec[a+b*x])^(m+n)/(d*Csc[a+b*x])^m,x] /;
FreeQ[{a,b,c,d,m,n},x] && KnownSecantIntegrandQ[u,x] && Not[IntegerQ[m]]
```

2: $\int u\,(c\,\mathrm{Tan}[a+b\,x])^m\,(d\,\mathrm{Csc}[a+b\,x])^n\,dx$ when KnownSecantIntegrandQ[u, x] $\wedge\ m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,\mathrm{Tan}[a+b\,x])^m\,(d\,\mathrm{Csc}[a+b\,x])^m}{(d\,\mathrm{Sec}[a+b\,x])^m} == 0$

Rule: If KnownSecantIntegrandQ[u, x] $\wedge\ m\notin\mathbb{Z}$**, then**

$$\int u\,(c\,\mathrm{Tan}[a+b\,x])^m\,(d\,\mathrm{Csc}[a+b\,x])^n\,dx \longrightarrow \frac{(c\,\mathrm{Tan}[a+b\,x])^m\,(d\,\mathrm{Csc}[a+b\,x])^m}{(d\,\mathrm{Sec}[a+b\,x])^m}\int \frac{u\,(d\,\mathrm{Sec}[a+b\,x])^{m}}{(d\,\mathrm{Csc}[a+b\,x])^{m-n}}\,dx$$

Program code:

```
Int[u_*(c_.*tan[a_.+b_.*x_])^m_.*(d_.*csc[a_.+b_.*x_])^n_.,x_Symbol] :=
  (c*Tan[a+b*x])^m*(d*Csc[a+b*x])^m/(d*Sec[a+b*x])^m*Int[ActivateTrig[u]*(d*Sec[a+b*x])^m/(d*Csc[a+b*x])^(m-n),x] /;
FreeQ[{a,b,c,d,m,n},x] && KnownSecantIntegrandQ[u,x] && Not[IntegerQ[m]]
```

4. $\int u\,(c\,\mathrm{Cot}[a+b\,x])^m\,(d\,\mathrm{Trig}[a+b\,x])^n\,dx$ when **KnownSecantIntegrandQ[u, x]**

1: $\int u\,(c\,\mathrm{Cot}[a+b\,x])^m\,(d\,\mathrm{Sec}[a+b\,x])^n\,dx$ when **KnownSecantIntegrandQ[u, x]** $\land\ m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,\mathrm{Cot}[a+b\,x])^m\,(d\,\mathrm{Sec}[a+b\,x])^m}{(d\,\mathrm{Csc}[a+b\,x])^m} == 0$

Rule: If KnownSecantIntegrandQ[u, x] $\land\ m \notin \mathbb{Z}$**, then**

$$\int u\,(c\,\mathrm{Cot}[a+b\,x])^m\,(d\,\mathrm{Sec}[a+b\,x])^n\,dx \longrightarrow \frac{(c\,\mathrm{Cot}[a+b\,x])^m\,(d\,\mathrm{Sec}[a+b\,x])^m}{(d\,\mathrm{Csc}[a+b\,x])^m}\int \frac{u\,(d\,\mathrm{Csc}[a+b\,x])^m}{(d\,\mathrm{Sec}[a+b\,x])^{m-n}}\,dx$$

Program code:

```
Int[u_*(c_.*cot[a_.+b_.*x_])^m_.*(d_.*sec[a_.+b_.*x_])^n_.,x_Symbol] :=
  (c*Cot[a+b*x])^m*(d*Sec[a+b*x])^m/(d*Csc[a+b*x])^m*Int[ActivateTrig[u]*(d*Csc[a+b*x])^m/(d*Sec[a+b*x])^(m-n),x] /;
FreeQ[{a,b,c,d,m,n},x] && KnownSecantIntegrandQ[u,x] && Not[IntegerQ[m]]
```

2: $\int u\,(c\,\mathrm{Cot}[a+b\,x])^m\,(d\,\mathrm{Csc}[a+b\,x])^n\,dx$ when **KnownSecantIntegrandQ[u, x]** $\land\ m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,\mathrm{Cot}[a+b\,x])^m\,(d\,\mathrm{Sec}[a+b\,x])^m}{(d\,\mathrm{Csc}[a+b\,x])^m} == 0$

Rule: If KnownSecantIntegrandQ[u, x] $\land\ m \notin \mathbb{Z}$**, then**

$$\int u\,(c\,\mathrm{Cot}[a+b\,x])^m\,(d\,\mathrm{Csc}[a+b\,x])^n\,dx \longrightarrow \frac{(c\,\mathrm{Cot}[a+b\,x])^m\,(d\,\mathrm{Sec}[a+b\,x])^m}{(d\,\mathrm{Csc}[a+b\,x])^m}\int \frac{u\,(d\,\mathrm{Csc}[a+b\,x])^{m+n}}{(d\,\mathrm{Sec}[a+b\,x])^m}\,dx$$

Program code:

```
Int[u_*(c_.*cot[a_.+b_.*x_])^m_.*(d_.*csc[a_.+b_.*x_])^n_.,x_Symbol] :=
  (c*Cot[a+b*x])^m*(d*Sec[a+b*x])^m/(d*Csc[a+b*x])^m*Int[ActivateTrig[u]*(d*Csc[a+b*x])^(m+n)/(d*Sec[a+b*x])^m,x] /;
FreeQ[{a,b,c,d,m,n},x] && KnownSecantIntegrandQ[u,x] && Not[IntegerQ[m]]
```

2. $\int u\,(c\,\mathrm{Trig}[a+b\,x])^m\,dx$ when $m\notin\mathbb{Z}\,\wedge$ KnownSecantIntegrandQ[u, x]

1: $\int u\,(c\,\mathrm{Sin}[a+b\,x])^m\,dx$ when $m\notin\mathbb{Z}\,\wedge$ KnownSecantIntegrandQ[u, x]

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((c\,\mathrm{Csc}[a+b\,x])^m\,(c\,\mathrm{Sin}[a+b\,x])^m\right)=0$

Rule: If $m\notin\mathbb{Z}\,\wedge$ KnownSecantIntegrandQ[u, x], **then**

$$\int u\,(c\,\mathrm{Sin}[a+b\,x])^m\,dx \longrightarrow (c\,\mathrm{Csc}[a+b\,x])^m\,(c\,\mathrm{Sin}[a+b\,x])^m\int\frac{u}{(c\,\mathrm{Csc}[a+b\,x])^m}\,dx$$

Program code:

```
Int[u_*(c_.*sin[a_.+b_.*x_])^m_.,x_Symbol] :=
  (c*Csc[a+b*x])^m*(c*Sin[a+b*x])^m*Int[ActivateTrig[u]/(c*Csc[a+b*x])^m,x] /;
FreeQ[{a,b,c,m},x] && Not[IntegerQ[m]] && KnownSecantIntegrandQ[u,x]
```

2: $\int u\,(c\,\mathrm{Cos}[a+b\,x])^m\,dx$ when $m\notin\mathbb{Z}\,\wedge$ KnownSecantIntegrandQ[u, x]

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((c\,\mathrm{Cos}[a+b\,x])^m\,(c\,\mathrm{Sec}[a+b\,x])^m\right)=0$

Rule: If $m\notin\mathbb{Z}\,\wedge$ KnownSecantIntegrandQ[u, x], **then**

$$\int u\,(c\,\mathrm{Cos}[a+b\,x])^m\,dx \longrightarrow (c\,\mathrm{Cos}[a+b\,x])^m\,(c\,\mathrm{Sec}[a+b\,x])^m\int\frac{u}{(c\,\mathrm{Sec}[a+b\,x])^m}\,dx$$

Program code:

```
Int[u_*(c_.*cos[a_.+b_.*x_])^m_.,x_Symbol] :=
  (c*Cos[a+b*x])^m*(c*Sec[a+b*x])^m*Int[ActivateTrig[u]/(c*Sec[a+b*x])^m,x] /;
FreeQ[{a,b,c,m},x] && Not[IntegerQ[m]] && KnownSecantIntegrandQ[u,x]
```

3: $\int u\,(c\,\mathrm{Tan}[a+b\,x])^m\,dx$ when $m\notin\mathbb{Z}\,\land$ KnownSecantIntegrandQ[u, x]

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(c\,\mathrm{Tan}[a+b\,x])^m\,(c\,\mathrm{Csc}[a+b\,x])^m}{(c\,\mathrm{Sec}[a+b\,x])^m} == 0$

- **Rule: If** $m\notin\mathbb{Z}\,\land$ KnownSecantIntegrandQ[u, x], **then**

$$\int u\,(c\,\mathrm{Tan}[a+b\,x])^m\,dx \longrightarrow \frac{(c\,\mathrm{Tan}[a+b\,x])^m\,(c\,\mathrm{Csc}[a+b\,x])^m}{(c\,\mathrm{Sec}[a+b\,x])^m}\int \frac{u\,(c\,\mathrm{Sec}[a+b\,x])^m}{(c\,\mathrm{Csc}[a+b\,x])^m}\,dx$$

Program code:

```
Int[u_*(c_.*tan[a_.+b_.*x_])^m_.,x_Symbol] :=
  (c*Tan[a+b*x])^m*(c*Csc[a+b*x])^m/(c*Sec[a+b*x])^m*Int[ActivateTrig[u]*(c*Sec[a+b*x])^m/(c*Csc[a+b*x])^m,x] /;
FreeQ[{a,b,c,m},x] && Not[IntegerQ[m]] && KnownSecantIntegrandQ[u,x]
```

4: $\int u\,(c\,\mathrm{Cot}[a+b\,x])^m\,dx$ when $m\notin\mathbb{Z}\,\land$ KnownSecantIntegrandQ[u, x]

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(c\,\mathrm{Cot}[a+b\,x])^m\,(c\,\mathrm{Sec}[a+b\,x])^m}{(c\,\mathrm{Csc}[a+b\,x])^m} == 0$

- **Rule: If** $m\notin\mathbb{Z}\,\land$ KnownSecantIntegrandQ[u, x], **then**

$$\int u\,(c\,\mathrm{Cot}[a+b\,x])^m\,dx \longrightarrow \frac{(c\,\mathrm{Cot}[a+b\,x])^m\,(c\,\mathrm{Sec}[a+b\,x])^m}{(c\,\mathrm{Csc}[a+b\,x])^m}\int \frac{u\,(c\,\mathrm{Csc}[a+b\,x])^m}{(c\,\mathrm{Sec}[a+b\,x])^m}\,dx$$

Program code:

```
Int[u_*(c_.*cot[a_.+b_.*x_])^m_.,x_Symbol] :=
  (c*Cot[a+b*x])^m*(c*Sec[a+b*x])^m/(c*Csc[a+b*x])^m*Int[ActivateTrig[u]*(c*Csc[a+b*x])^m/(c*Sec[a+b*x])^m,x] /;
FreeQ[{a,b,c,m},x] && Not[IntegerQ[m]] && KnownSecantIntegrandQ[u,x]
```

3. $\int u\,(A+B\,\text{Cos}[a+b\,x])\,dx$ when KnownSecantIntegrandQ[u, x]

1: $\int u\,(c\,\text{Sec}[a+b\,x])^n\,(A+B\,\text{Cos}[a+b\,x])\,dx$ when KnownSecantIntegrandQ[u, x]

Derivation: Algebraic normalization

Rule: If KnownSecantIntegrandQ[u, x], then

$$\int u\,(c\,\text{Sec}[a+b\,x])^n\,(A+B\,\text{Cos}[a+b\,x])\,dx\;\longrightarrow\;c\int u\,(c\,\text{Sec}[a+b\,x])^{n-1}\,(B+A\,\text{Sec}[a+b\,x])\,dx$$

Program code:

```
Int[u_*(c_.*sec[a_.+b_.*x_])^n_.*(A_+B_.*cos[a_.+b_.*x_]),x_Symbol] :=
  c*Int[ActivateTrig[u]*(c*Sec[a+b*x])^(n-1)*(B+A*Sec[a+b*x]),x] /;
FreeQ[{a,b,c,A,B,n},x] && KnownSecantIntegrandQ[u,x]
```

```
Int[u_*(c_.*csc[a_.+b_.*x_])^n_.*(A_+B_.*sin[a_.+b_.*x_]),x_Symbol] :=
  c*Int[ActivateTrig[u]*(c*Csc[a+b*x])^(n-1)*(B+A*Csc[a+b*x]),x] /;
FreeQ[{a,b,c,A,B,n},x] && KnownSecantIntegrandQ[u,x]
```

2: $\int u\,(A+B\,\text{Cos}[a+b\,x])\,dx$ when KnownSecantIntegrandQ[u, x]

Derivation: Algebraic normalization

Rule: If KnownSecantIntegrandQ[u, x], then

$$\int u\,(A+B\,\text{Cos}[a+b\,x])\,dx\;\longrightarrow\;\int\frac{u\,(B+A\,\text{Sec}[a+b\,x])}{\text{Sec}[a+b\,x]}\,dx$$

Program code:

```
Int[u_*(A_+B_.*cos[a_.+b_.*x_]),x_Symbol] :=
  Int[ActivateTrig[u]*(B+A*Sec[a+b*x])/Sec[a+b*x],x] /;
FreeQ[{a,b,A,B},x] && KnownSecantIntegrandQ[u,x]
```

```
Int[u_*(A_+B_.*sin[a_.+b_.*x_]),x_Symbol] :=
  Int[ActivateTrig[u]*(B+A*Csc[a+b*x])/Csc[a+b*x],x] /;
FreeQ[{a,b,A,B},x] && KnownSecantIntegrandQ[u,x]
```

4. $\int u\,(A+B\,\text{Cos}[a+b\,x]+C\,\text{Cos}[a+b\,x]^2)\,dx$ when **KnownSecantIntegrandQ[u, x]**

1: $\int u\,(c\,\text{Sec}[a+b\,x])^n\,(A+B\,\text{Cos}[a+b\,x]+C\,\text{Cos}[a+b\,x]^2)\,dx$ when **KnownSecantIntegrandQ[u, x]**

Derivation: Algebraic normalization

Rule: If KnownSecantIntegrandQ[u, x], then

$$\int u\,(c\,\text{Sec}[a+b\,x])^n\,(A+B\,\text{Cos}[a+b\,x]+C\,\text{Cos}[a+b\,x]^2)\,dx \longrightarrow c^2\int u\,(c\,\text{Sec}[a+b\,x])^{n-2}\,(C+B\,\text{Sec}[a+b\,x]+A\,\text{Sec}[a+b\,x]^2)\,dx$$

Program code:

```
Int[u_.*(c_.*sec[a_.+b_.*x_])^n_.*(A_.+B_.*cos[a_.+b_.*x_]+C_.*cos[a_.+b_.*x_]^2),x_Symbol] :=
  c^2*Int[ActivateTrig[u]*(c*Sec[a+b*x])^(n-2)*(C+B*Sec[a+b*x]+A*Sec[a+b*x]^2),x] /;
FreeQ[{a,b,c,A,B,C,n},x] && KnownSecantIntegrandQ[u,x]
```

```
Int[u_.*(c_.*csc[a_.+b_.*x_])^n_.*(A_.+B_.*sin[a_.+b_.*x_]+C_.*sin[a_.+b_.*x_]^2),x_Symbol] :=
  c^2*Int[ActivateTrig[u]*(c*Csc[a+b*x])^(n-2)*(C+B*Csc[a+b*x]+A*Csc[a+b*x]^2),x] /;
FreeQ[{a,b,c,A,B,C,n},x] && KnownSecantIntegrandQ[u,x]
```

```
Int[u_.*(c_.*sec[a_.+b_.*x_])^n_.*(A_+C_.*cos[a_.+b_.*x_]^2),x_Symbol] :=
  c^2*Int[ActivateTrig[u]*(c*Sec[a+b*x])^(n-2)*(C+A*Sec[a+b*x]^2),x] /;
FreeQ[{a,b,c,A,C,n},x] && KnownSecantIntegrandQ[u,x]
```

```
Int[u_.*(c_.*csc[a_.+b_.*x_])^n_.*(A_+C_.*sin[a_.+b_.*x_]^2),x_Symbol] :=
  c^2*Int[ActivateTrig[u]*(c*Csc[a+b*x])^(n-2)*(C+A*Csc[a+b*x]^2),x] /;
FreeQ[{a,b,c,A,C,n},x] && KnownSecantIntegrandQ[u,x]
```

2: $\int u\left(A+B\,\mathrm{Cos}[a+b\,x]+C\,\mathrm{Cos}[a+b\,x]^2\right)dx$ when KnownSecantIntegrandQ[u, x]

Derivation: Algebraic normalization

Rule: If KnownSecantIntegrandQ [u, x], then

$$\int u\left(A+B\,\mathrm{Cos}[a+b\,x]+C\,\mathrm{Cos}[a+b\,x]^2\right)dx\;\longrightarrow\;\int\frac{u\left(C+B\,\mathrm{Sec}[a+b\,x]+A\,\mathrm{Sec}[a+b\,x]^2\right)}{\mathrm{Sec}[a+b\,x]^2}\,dx$$

Program code:

```
Int[u_*(A_.+B_.*cos[a_.+b_.*x_]+C_.*cos[a_.+b_.*x_]^2),x_Symbol] :=
  Int[ActivateTrig[u]*(C+B*Sec[a+b*x]+A*Sec[a+b*x]^2)/Sec[a+b*x]^2,x] /;
FreeQ[{a,b,A,B,C},x] && KnownSecantIntegrandQ[u,x]
```

```
Int[u_*(A_.+B_.*sin[a_.+b_.*x_]+C_.*sin[a_.+b_.*x_]^2),x_Symbol] :=
  Int[ActivateTrig[u]*(C+B*Csc[a+b*x]+A*Csc[a+b*x]^2)/Csc[a+b*x]^2,x] /;
FreeQ[{a,b,A,B,C},x] && KnownSecantIntegrandQ[u,x]
```

```
Int[u_*(A_+C_.*cos[a_.+b_.*x_]^2),x_Symbol] :=
  Int[ActivateTrig[u]*(C+A*Sec[a+b*x]^2)/Sec[a+b*x]^2,x] /;
FreeQ[{a,b,A,C},x] && KnownSecantIntegrandQ[u,x]
```

```
Int[u_*(A_+C_.*sin[a_.+b_.*x_]^2),x_Symbol] :=
  Int[ActivateTrig[u]*(C+A*Csc[a+b*x]^2)/Csc[a+b*x]^2,x] /;
FreeQ[{a,b,A,C},x] && KnownSecantIntegrandQ[u,x]
```

5: $\int u\left(A\,\mathrm{Sec}[a+b\,x]^n+B\,\mathrm{Sec}[a+b\,x]^{n+1}+C\,\mathrm{Sec}[a+b\,x]^{n+2}\right)dx$

Derivation: Algebraic normalization

Rule:

$$\int u\left(A\,\mathrm{Sec}[a+b\,x]^n+B\,\mathrm{Sec}[a+b\,x]^{n+1}+C\,\mathrm{Sec}[a+b\,x]^{n+2}\right)dx\;\longrightarrow\;\int u\,\mathrm{Sec}[a+b\,x]^n\left(A+B\,\mathrm{Sec}[a+b\,x]+C\,\mathrm{Sec}[a+b\,x]^2\right)dx$$

Program code:

```
Int[u_*(A_.*sec[a_.+b_.*x_]^n_.+B_.*sec[a_.+b_.*x_]^n1_+C_.*sec[a_.+b_.*x_]^n2_),x_Symbol] :=
  Int[ActivateTrig[u]*Sec[a+b*x]^n*(A+B*Sec[a+b*x]+C*Sec[a+b*x]^2),x] /;
FreeQ[{a,b,A,B,C,n},x] && EqQ[n1,n+1] && EqQ[n2,n+2]
```

```
Int[u_*(A_.*csc[a_.+b_.*x_]^n_.+B_.*csc[a_.+b_.*x_]^n1_+C_.*csc[a_.+b_.*x_]^n2_),x_Symbol] :=
  Int[ActivateTrig[u]*Csc[a+b*x]^n*(A+B*Csc[a+b*x]+C*Csc[a+b*x]^2),x] /;
FreeQ[{a,b,A,B,C,n},x] && EqQ[n1,n+1] && EqQ[n2,n+2]
```