Rules for integrands of the form $(a\,\mathrm{Csc}[e+f\,x])^m\,(b\,\mathrm{Sec}[e+f\,x])^n$

1: $\int (a\,\mathrm{Csc}[e+f\,x])^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $m+n-2=0\ \wedge\ n\neq 1$

- **Reference: G&R 2.510.3, CRC 334a, A&S 4.3.128b with $m+n-2=0$**
- **Reference: G&R 2.510.6, CRC 334b, A&S 4.3.128a with $m+n-2=0$**
- **Rule: If $m+n-2=0\ \wedge\ n\neq 1$, then**

$$\int (a\,\mathrm{Csc}[e+f\,x])^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx \longrightarrow \frac{a\,b\,(a\,\mathrm{Csc}[e+f\,x])^{m-1}\,(b\,\mathrm{Sec}[e+f\,x])^{n-1}}{f\,(n-1)}$$

- **Program code:**

```
Int[(a_.*csc[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  a*b*(a*Csc[e+f*x])^(m-1)*(b*Sec[e+f*x])^(n-1)/(f*(n-1)) /;
FreeQ[{a,b,e,f,m,n},x] && EqQ[m+n-2,0] && NeQ[n,1]
```

2: $\int \mathrm{Csc}[e+f\,x]^m\,\mathrm{Sec}[e+f\,x]^n\,dx$ when $\left(m\mid n\mid \frac{m+n}{2}\right)\in\mathbb{Z}$

- **Derivation: Integration by substitution**
- **Basis: If $\left(m\mid n\mid \frac{m+n}{2}\right)\in\mathbb{Z}$, then $\mathrm{Csc}[e+f\,x]^m\,\mathrm{Sec}[e+f\,x]^n = \frac{1}{f}\,\mathrm{Subst}\left[\frac{(1+x^2)^{\frac{m+n}{2}-1}}{x^m},\ x,\ \mathrm{Tan}[e+f\,x]\right]\,\partial_x \mathrm{Tan}[e+f\,x]$**
- **Rule: If $\left(m\mid n\mid \frac{m+n}{2}\right)\in\mathbb{Z}$, then**

$$\int \mathrm{Csc}[e+f\,x]^m\,\mathrm{Sec}[e+f\,x]^n\,dx \longrightarrow \frac{1}{f}\,\mathrm{Subst}\left[\int \frac{\left(1+x^2\right)^{\frac{m+n}{2}-1}}{x^m}\,dx,\ x,\ \mathrm{Tan}[e+f\,x]\right]$$

- **Program code:**

```
Int[csc[e_.+f_.*x_]^m_.*sec[e_.+f_.*x_]^n_.,x_Symbol] :=
  1/f*Subst[Int[(1+x^2)^((m+n)/2-1)/x^m,x],x,Tan[e+f*x]] /;
FreeQ[{e,f},x] && IntegersQ[m,n,(m+n)/2]
```

3: $\int (a \operatorname{Csc}[e+f x])^m \operatorname{Sec}[e+f x]^n \, dx$ when $\frac{n+1}{2} \in \mathbb{Z}$

- **Derivation: Integration by substitution**
- **Basis: If** $\frac{n-1}{2} \in \mathbb{Z}$**, then** $(a \operatorname{Csc}[e+f x])^m \operatorname{Sec}[e+f x]^n == -\frac{1}{f a^n} \operatorname{Subst}\left[\frac{x^{m+n-1}}{\left(-1+\frac{x^2}{a^2}\right)^{\frac{n+1}{2}}}, x, a \operatorname{Csc}[e+f x]\right] \partial_x (a \operatorname{Csc}[e+f x])$
- **Rule: If** $\frac{n+1}{2} \in \mathbb{Z}$**, then**

$$\int (a \operatorname{Csc}[e+f x])^m \operatorname{Sec}[e+f x]^n \, dx \longrightarrow -\frac{1}{f a^n} \operatorname{Subst}\left[\int \frac{x^{m+n-1}}{\left(-1+\frac{x^2}{a^2}\right)^{\frac{n+1}{2}}} dx, x, a \operatorname{Csc}[e+f x]\right]$$

Program code:

```
Int[(a_.*csc[e_.+f_.*x_])^m_*sec[e_.+f_.*x_]^n_.,x_Symbol] :=
  -1/(f*a^n)*Subst[Int[x^(m+n-1)/(-1+x^2/a^2)^((n+1)/2),x],x,a*Csc[e+f*x]] /;
FreeQ[{a,e,f,m},x] && IntegerQ[(n+1)/2] && Not[IntegerQ[(m+1)/2] && LtQ[0,m,n]]
```

```
Int[(a_.*sec[e_.+f_.*x_])^m_*csc[e_.+f_.*x_]^n_.,x_Symbol] :=
  1/(f*a^n)*Subst[Int[x^(m+n-1)/(-1+x^2/a^2)^((n+1)/2),x],x,a*Sec[e+f*x]] /;
FreeQ[{a,e,f,m},x] && IntegerQ[(n+1)/2] && Not[IntegerQ[(m+1)/2] && LtQ[0,m,n]]
```

4. $\int (a\,\mathrm{Csc}[e+f\,x])^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $m>1$

1: $\int (a\,\mathrm{Csc}[e+f\,x])^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $m>1\,\wedge\,n<-1$

Reference: G&R 2.510.1

Reference: G&R 2.510.4

Rule: If $m>1\,\wedge\,n<-1$, then

$$\int (a\,\mathrm{Csc}[e+f\,x])^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx\;\longrightarrow$$
$$-\frac{a\,(a\,\mathrm{Csc}[e+f\,x])^{m-1}\,(b\,\mathrm{Sec}[e+f\,x])^{n+1}}{f\,b\,(m-1)}+\frac{a^2\,(n+1)}{b^2\,(m-1)}\int (a\,\mathrm{Csc}[e+f\,x])^{m-2}\,(b\,\mathrm{Sec}[e+f\,x])^{n+2}\,dx$$

Program code:

```
Int[(a_.*csc[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  -a*(a*Csc[e+f*x])^(m-1)*(b*Sec[e+f*x])^(n+1)/(f*b*(m-1)) +
  a^2*(n+1)/(b^2*(m-1))*Int[(a*Csc[e+f*x])^(m-2)*(b*Sec[e+f*x])^(n+2),x] /;
FreeQ[{a,b,e,f},x] && GtQ[m,1] && LtQ[n,-1] && IntegersQ[2*m,2*n]
```

```
Int[(a_.*csc[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(a*Csc[e+f*x])^(m+1)*(b*Sec[e+f*x])^(n-1)/(f*a*(n-1)) +
  b^2*(m+1)/(a^2*(n-1))*Int[(a*Csc[e+f*x])^(m+2)*(b*Sec[e+f*x])^(n-2),x] /;
FreeQ[{a,b,e,f},x] && GtQ[n,1] && LtQ[m,-1] && IntegersQ[2*m,2*n]
```

2: $\int (a\,Csc[e+f\,x])^m\,(b\,Sec[e+f\,x])^n\,dx$ when $m>1$

Reference: G&R 2.510.2, CRC 323b, A&S 4.3.127b

Reference: G&R 2.510.5, CRC 323a, A&S 4.3.127a

Rule: If $m>1$, then

$$\int (a\,Csc[e+f\,x])^m\,(b\,Sec[e+f\,x])^n\,dx \longrightarrow$$
$$-\frac{a\,b\,(a\,Csc[e+f\,x])^{m-1}\,(b\,Sec[e+f\,x])^{n-1}}{f\,(m-1)}+\frac{a^2\,(m+n-2)}{m-1}\int (a\,Csc[e+f\,x])^{m-2}\,(b\,Sec[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_.*csc[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_.,x_Symbol] :=
  -a*b*(a*Csc[e+f*x])^(m-1)*(b*Sec[e+f*x])^(n-1)/(f*(m-1)) +
  a^2*(m+n-2)/(m-1)*Int[(a*Csc[e+f*x])^(m-2)*(b*Sec[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,n},x] && GtQ[m,1] && IntegersQ[2*m,2*n] && Not[GtQ[n,m]]
```

```
Int[(a_.*csc[e_.+f_.*x_])^m_.*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  a*b*(a*Csc[e+f*x])^(m-1)*(b*Sec[e+f*x])^(n-1)/(f*(n-1)) +
  b^2*(m+n-2)/(n-1)*Int[(a*Csc[e+f*x])^m*(b*Sec[e+f*x])^(n-2),x] /;
FreeQ[{a,b,e,f,m},x] && GtQ[n,1] && IntegersQ[2*m,2*n]
```

5: $\int (a\,Csc[e+f\,x])^m\,(b\,Sec[e+f\,x])^n\,dx$ when $m<-1\,\wedge\,m+n\neq 0$

Reference: G&R 2.510.3, CRC 334a, A&S 4.3.128b

Reference: G&R 2.510.6, CRC 334b, A&S 4.3.128a

Rule: If $m<-1\,\wedge\,m+n\neq 0$, then

$$\int (a\,Csc[e+f\,x])^m\,(b\,Sec[e+f\,x])^n\,dx \longrightarrow \frac{b\,(a\,Csc[e+f\,x])^{m+1}\,(b\,Sec[e+f\,x])^{n-1}}{a\,f\,(m+n)}+\frac{m+1}{a^2\,(m+n)}\int (a\,Csc[e+f\,x])^{m+2}\,(b\,Sec[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_.*csc[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_.,x_Symbol] :=
  b*(a*Csc[e+f*x])^(m+1)*(b*Sec[e+f*x])^(n-1)/(a*f*(m+n)) +
  (m+1)/(a^2*(m+n))*Int[(a*Csc[e+f*x])^(m+2)*(b*Sec[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,n},x] && LtQ[m,-1] && NeQ[m+n,0] && IntegersQ[2*m,2*n]
```

```
Int[(a_.*csc[e_.+f_.*x_])^m_.*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  -a*(a*Csc[e+f*x])^(m-1)*(b*Sec[e+f*x])^(n+1)/(b*f*(m+n)) +
  (n+1)/(b^2*(m+n))*Int[(a*Csc[e+f*x])^m*(b*Sec[e+f*x])^(n+2),x] /;
FreeQ[{a,b,e,f,m},x] && LtQ[n,-1] && NeQ[m+n,0] && IntegersQ[2*m,2*n]
```

6: $\int (a\,\mathrm{Csc}[e+f\,x])^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $n\notin\mathbb{Z}\,\land\,m+n==0$

Derivation: Piecewise constant extraction

Basis: If $m+n==0$**, then** $\partial_x \frac{(a\,\mathrm{Csc}[e+f\,x])^m\,(b\,\mathrm{Sec}[e+f\,x])^n}{\mathrm{Tan}[e+f\,x]^n}==0$

Rule: If $n\notin\mathbb{Z}\,\land\,m+n==0$**, then**

$$\int (a\,\mathrm{Csc}[e+f\,x])^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx \longrightarrow \frac{(a\,\mathrm{Csc}[e+f\,x])^m\,(b\,\mathrm{Sec}[e+f\,x])^n}{\mathrm{Tan}[e+f\,x]^n}\int \mathrm{Tan}[e+f\,x]^n\,dx$$

Program code:

```
Int[(a_.*csc[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  (a*Csc[e+f*x])^m*(b*Sec[e+f*x])^n/Tan[e+f*x]^n*Int[Tan[e+f*x]^n,x] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[n]] && EqQ[m+n,0]
```

7. $\int (a \operatorname{Csc}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx$

1: $\int (a \operatorname{Csc}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx$ when $m-\frac{1}{2} \in \mathbb{Z} \bigwedge n-\frac{1}{2} \in \mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \left((a \operatorname{Csc}[e+f x])^m (b \operatorname{Sec}[e+f x])^n (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n\right) == 0$
- **Rule: If** $m-\frac{1}{2} \in \mathbb{Z} \bigwedge n-\frac{1}{2} \in \mathbb{Z}$**, then**

$$\int (a \operatorname{Csc}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx \longrightarrow$$

$$(a \operatorname{Csc}[e+f x])^m (b \operatorname{Sec}[e+f x])^n (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \int (a \operatorname{Sin}[e+f x])^{-m} (b \operatorname{Cos}[e+f x])^{-n} \, dx$$

- **Program code:**

```
Int[(a_.*csc[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  (a*Csc[e+f*x])^m*(b*Sec[e+f*x])^n*(a*Sin[e+f*x])^m*(b*Cos[e+f*x])^n*Int[(a*Sin[e+f*x])^(-m)*(b*Cos[e+f*x])^(-n),x] /;
FreeQ[{a,b,e,f,m,n},x] && IntegerQ[m-1/2] && IntegerQ[n-1/2]
```

2: $\int (a \operatorname{Csc}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \left((a \operatorname{Csc}[e+f x])^m (b \operatorname{Sec}[e+f x])^n (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n\right) == 0$
- **Rule:**

$$\int (a \operatorname{Csc}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx \longrightarrow$$

$$\frac{a^2}{b^2} (a \operatorname{Csc}[e+f x])^{m-1} (b \operatorname{Sec}[e+f x])^{n+1} (a \operatorname{Sin}[e+f x])^{m-1} (b \operatorname{Cos}[e+f x])^{n+1} \int (a \operatorname{Sin}[e+f x])^{-m} (b \operatorname{Cos}[e+f x])^{-n} \, dx$$

- **Program code:**

```
Int[(a_.*csc[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  a^2/b^2*(a*Csc[e+f*x])^(m-1)*(b*Sec[e+f*x])^(n+1)*(a*Sin[e+f*x])^(m-1)*(b*Cos[e+f*x])^(n+1)*
    Int[(a*Sin[e+f*x])^(-m)*(b*Cos[e+f*x])^(-n),x] /;
FreeQ[{a,b,e,f,m,n},x] && Not[SimplerQ[-m,-n]]
```

```
Int[(a_.*sec[e_.+f_.*x_])^m_*(b_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  a^2/b^2*(a*Sec[e+f*x])^(m-1)*(b*Csc[e+f*x])^(n+1)*(a*Cos[e+f*x])^(m-1)*(b*Sin[e+f*x])^(n+1)*
    Int[(a*Cos[e+f*x])^(-m)*(b*Sin[e+f*x])^(-n),x] /;
FreeQ[{a,b,e,f,m,n},x]
```