Rules for integrands involving hyperbolic integral functions

1. $\int u\, \text{SinhIntegral}[a + b\, x]\, dx$

1: $\int \text{SinhIntegral}[a + b\, x]\, dx$

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{SinhIntegral}[a + b\, x]\, dx \longrightarrow \frac{(a + b\, x)\, \text{SinhIntegral}[a + b\, x]}{b} - \frac{\text{Cosh}[a + b\, x]}{b}$$

- **Program code:**

```
Int[SinhIntegral[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*SinhIntegral[a+b*x]/b - Cosh[a+b*x]/b/;
FreeQ[{a,b},x]
```

```
Int[CoshIntegral[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*CoshIntegral[a+b*x]/b - Sinh[a+b*x]/b /;
FreeQ[{a,b},x]
```

2. $\int (c + d\,x)^m\,\text{SinhIntegral}[a + b\,x]\,dx$

1: $\int \frac{\text{SinhIntegral}[b\,x]}{x}\,dx$

- Basis: $\text{SinhIntegral}[z] == -\frac{1}{2}\left(\text{ExpIntegralE}[1, -z] - \text{ExpIntegralE}[1, z] + \text{Log}[-z] - \text{Log}[z]\right)$

- Basis: $\text{CoshIntegral}[z] == -\frac{1}{2}\left(\text{ExpIntegralE}[1, -z] + \text{ExpIntegralE}[1, z] + \text{Log}[-z] - \text{Log}[z]\right)$

- Rule:

$$\int \frac{\text{SinhIntegral}[b\,x]}{x}\,dx \longrightarrow$$
$$\frac{1}{2}\,b\,x\,\text{HypergeometricPFQ}[\{1, 1, 1\}, \{2, 2, 2\}, -b\,x] + \frac{1}{2}\,b\,x\,\text{HypergeometricPFQ}[\{1, 1, 1\}, \{2, 2, 2\}, b\,x]$$

- Program code:

```
Int[SinhIntegral[b_.*x_]/x_,x_Symbol] :=
  1/2*b*x*HypergeometricPFQ[{1,1,1},{2,2,2},-b*x] +
  1/2*b*x*HypergeometricPFQ[{1,1,1},{2,2,2},b*x] /;
FreeQ[b,x]
```

```
Int[CoshIntegral[b_.*x_]/x_,x_Symbol] :=
  -1/2*b*x*HypergeometricPFQ[{1,1,1},{2,2,2},-b*x] +
  1/2*b*x*HypergeometricPFQ[{1,1,1},{2,2,2},b*x] +
  EulerGamma*Log[x] +
  1/2*Log[b*x]^2 /;
FreeQ[b,x]
```

2: $\int (c + d x)^m \text{SinhIntegral}[a + b x] \, dx$ when $m \neq -1$

Derivation: Integration by parts

Rule: If $m \neq -1$, then

$$\int (c + d x)^m \text{SinhIntegral}[a + b x] \, dx \longrightarrow \frac{(c + d x)^{m+1} \text{SinhIntegral}[a + b x]}{d (m + 1)} - \frac{b}{d (m + 1)} \int \frac{(c + d x)^{m+1} \text{Sinh}[a + b x]}{a + b x} \, dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*SinhIntegral[a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*SinhIntegral[a+b*x]/(d*(m+1)) -
  b/(d*(m+1))*Int[(c+d*x)^(m+1)*Sinh[a+b*x]/(a+b*x),x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1]
```

```
Int[(c_.+d_.*x_)^m_.*CoshIntegral[a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*CoshIntegral[a+b*x]/(d*(m+1)) -
  b/(d*(m+1))*Int[(c+d*x)^(m+1)*Cosh[a+b*x]/(a+b*x),x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1]
```

2. $\int u \, \text{SinhIntegral}[a + b x]^2 \, dx$

1: $\int \text{SinhIntegral}[a + b x]^2 \, dx$

Derivation: Integration by parts

Rule:

$$\int \text{SinhIntegral}[a + b x]^2 \, dx \longrightarrow \frac{(a + b x) \, \text{SinhIntegral}[a + b x]^2}{b} - 2 \int \text{Sinh}[a + b x] \, \text{SinhIntegral}[a + b x] \, dx$$

Program code:

```
Int[SinhIntegral[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*SinhIntegral[a+b*x]^2/b -
  2*Int[Sinh[a+b*x]*SinhIntegral[a+b*x],x] /;
FreeQ[{a,b},x]
```

```
Int[CoshIntegral[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*CoshIntegral[a+b*x]^2/b -
  2*Int[Cosh[a+b*x]*CoshIntegral[a+b*x],x] /;
FreeQ[{a,b},x]
```

2. $\int (c+d\,x)^m\,\text{SinhIntegral}[a+b\,x]^2\,dx$

1: $\int x^m\,\text{SinhIntegral}[b\,x]^2\,dx$ when $m\in\mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $m\in\mathbb{Z}^+$, then

$$\int x^m\,\text{SinhIntegral}[b\,x]^2\,dx \longrightarrow \frac{x^{m+1}\,\text{SinhIntegral}[b\,x]^2}{m+1}-\frac{2}{m+1}\int x^m\,\text{Sinh}[b\,x]\,\text{SinhIntegral}[b\,x]\,dx$$

Program code:

```
Int[x_^m_.*SinhIntegral[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*SinhIntegral[b*x]^2/(m+1) -
  2/(m+1)*Int[x^m*Sinh[b*x]*SinhIntegral[b*x],x] /;
FreeQ[b,x] && IGtQ[m,0]
```

```
Int[x_^m_.*CoshIntegral[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*CoshIntegral[b*x]^2/(m+1) -
  2/(m+1)*Int[x^m*Cosh[b*x]*CoshIntegral[b*x],x] /;
FreeQ[b,x] && IGtQ[m,0]
```

2: $\int (c+d\,x)^m\,\text{SinhIntegral}[a+b\,x]^2\,dx$ when $m\in\mathbb{Z}^+$

Derivation: Iterated integration by parts

Rule: If $m\in\mathbb{Z}^+$, then

$$\int (c+d\,x)^m\,\text{SinhIntegral}[a+b\,x]^2\,dx \longrightarrow$$

$$\frac{(a+b\,x)\,(c+d\,x)^m\,\text{SinhIntegral}[a+b\,x]^2}{b\,(m+1)}-$$

$$\frac{2}{m+1}\int(c+d\,x)^m\,\text{Sinh}[a+b\,x]\,\text{SinhIntegral}[a+b\,x]\,dx+\frac{(b\,c-a\,d)\,m}{b\,(m+1)}\int(c+d\,x)^{m-1}\,\text{SinhIntegral}[a+b\,x]^2\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*SinhIntegral[a_+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*(c+d*x)^m*SinhIntegral[a+b*x]^2/(b*(m+1)) -
  2/(m+1)*Int[(c+d*x)^m*Sinh[a+b*x]*SinhIntegral[a+b*x],x] +
  (b*c-a*d)*m/(b*(m+1))*Int[(c+d*x)^(m-1)*SinhIntegral[a+b*x]^2,x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_.*CoshIntegral[a_+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*(c+d*x)^m*CoshIntegral[a+b*x]^2/(b*(m+1)) -
  2/(m+1)*Int[(c+d*x)^m*Cosh[a+b*x]*CoshIntegral[a+b*x],x] +
  (b*c-a*d)*m/(b*(m+1))*Int[(c+d*x)^(m-1)*CoshIntegral[a+b*x]^2,x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

x: $\int x^m\,\text{SinhIntegral}[a+b\,x]^2\,dx$ when $m+2\in\mathbb{Z}^-$

Derivation: Inverted integration by parts

Rule: If $m+2\in\mathbb{Z}^-$, then

$$\int x^m\,\text{SinhIntegral}[a+b\,x]^2\,dx\longrightarrow\frac{b\,x^{m+2}\,\text{SinhIntegral}[a+b\,x]^2}{a\,(m+1)}+\frac{x^{m+1}\,\text{SinhIntegral}[a+b\,x]^2}{m+1}-$$
$$\frac{2\,b}{a\,(m+1)}\int x^{m+1}\,\text{Sinh}[a+b\,x]\,\text{SinhIntegral}[a+b\,x]\,dx-\frac{b\,(m+2)}{a\,(m+1)}\int x^{m+1}\,\text{SinhIntegral}[a+b\,x]^2\,dx$$

Program code:

```
(* Int[x_^m_.*SinhIntegral[a_+b_.*x_]^2,x_Symbol] :=
  b*x^(m+2)*SinhIntegral[a+b*x]^2/(a*(m+1)) +
  x^(m+1)*SinhIntegral[a+b*x]^2/(m+1) -
  2*b/(a*(m+1))*Int[x^(m+1)*Sinh[a+b*x]*SinhIntegral[a+b*x],x] -
  b*(m+2)/(a*(m+1))*Int[x^(m+1)*SinhIntegral[a+b*x]^2,x] /;
FreeQ[{a,b},x] && ILtQ[m,-2] *)
```

```
(* Int[x_^m_.*CoshIntegral[a_+b_.*x_]^2,x_Symbol] :=
  b*x^(m+2)*CoshIntegral[a+b*x]^2/(a*(m+1)) +
  x^(m+1)*CoshIntegral[a+b*x]^2/(m+1) -
  2*b/(a*(m+1))*Int[x^(m+1)*Cosh[a+b*x]*CoshIntegral[a+b*x],x] -
  b*(m+2)/(a*(m+1))*Int[x^(m+1)*CoshIntegral[a+b*x]^2,x] /;
FreeQ[{a,b},x] && ILtQ[m,-2] *)
```

3. $\int u\,\text{Sinh}[a+b\,x]\,\text{SinhIntegral}[c+d\,x]\,dx$

1: $\int \text{Sinh}[a+b\,x]\,\text{SinhIntegral}[c+d\,x]\,dx$

Derivation: Integration by parts

Rule:

$$\int \text{Sinh}[a+b\,x]\,\text{SinhIntegral}[c+d\,x]\,dx \longrightarrow \frac{\text{Cosh}[a+b\,x]\,\text{SinhIntegral}[c+d\,x]}{b}-\frac{d}{b}\int \frac{\text{Cosh}[a+b\,x]\,\text{Sinh}[c+d\,x]}{c+d\,x}\,dx$$

Program code:

```
Int[Sinh[a_.+b_.*x_]*SinhIntegral[c_.+d_.*x_],x_Symbol] :=
  Cosh[a+b*x]*SinhIntegral[c+d*x]/b -
  d/b*Int[Cosh[a+b*x]*Sinh[c+d*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d},x]
```

```
Int[Cosh[a_.+b_.*x_]*CoshIntegral[c_.+d_.*x_],x_Symbol] :=
  Sinh[a+b*x]*CoshIntegral[c+d*x]/b -
  d/b*Int[Sinh[a+b*x]*Cosh[c+d*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d},x]
```

2. $\int (e+f\,x)^m\,\text{Sinh}[a+b\,x]\,\text{SinhIntegral}[c+d\,x]\,dx$

1: $\int (e+f\,x)^m\,\text{Sinh}[a+b\,x]\,\text{SinhIntegral}[c+d\,x]\,dx$ when $m\in\mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $m\in\mathbb{Z}^+$, then

$$\int (e+f\,x)^m\,\text{Sinh}[a+b\,x]\,\text{SinhIntegral}[c+d\,x]\,dx \longrightarrow$$

$$\frac{(e+f x)^m \operatorname{Cosh}[a+b x] \operatorname{SinhIntegral}[c+d x]}{b}-$$
$$\frac{d}{b}\int\frac{(e+f x)^m \operatorname{Cosh}[a+b x] \operatorname{Sinh}[c+d x]}{c+d x}\,dx-\frac{f m}{b}\int(e+f x)^{m-1} \operatorname{Cosh}[a+b x] \operatorname{SinhIntegral}[c+d x]\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sinh[a_.+b_.*x_]*SinhIntegral[c_.+d_.*x_],x_Symbol] :=
  (e+f*x)^m*Cosh[a+b*x]*SinhIntegral[c+d*x]/b -
  d/b*Int[(e+f*x)^m*Cosh[a+b*x]*Sinh[c+d*x]/(c+d*x),x] -
  f*m/b*Int[(e+f*x)^(m-1)*Cosh[a+b*x]*SinhIntegral[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0]
```

```
Int[(e_.+f_.*x_)^m_.*Cosh[a_.+b_.*x_]*CoshIntegral[c_.+d_.*x_],x_Symbol] :=
  (e+f*x)^m*Sinh[a+b*x]*CoshIntegral[c+d*x]/b -
  d/b*Int[(e+f*x)^m*Sinh[a+b*x]*Cosh[c+d*x]/(c+d*x),x] -
  f*m/b*Int[(e+f*x)^(m-1)*Sinh[a+b*x]*CoshIntegral[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0]
```

2: $\int(e+f x)^m \operatorname{Sinh}[a+b x] \operatorname{SinhIntegral}[c+d x]\,dx$ when $m+1\in\mathbb{Z}^-$

Derivation: Inverted integration by parts

Rule: If $m+1\in\mathbb{Z}^-$, then

$$\int(e+f x)^m \operatorname{Sinh}[a+b x] \operatorname{SinhIntegral}[c+d x]\,dx\longrightarrow$$
$$\frac{(e+f x)^{m+1} \operatorname{Sinh}[a+b x] \operatorname{SinhIntegral}[c+d x]}{f(m+1)}-$$
$$\frac{d}{f(m+1)}\int\frac{(e+f x)^{m+1} \operatorname{Sinh}[a+b x] \operatorname{Sinh}[c+d x]}{c+d x}\,dx-\frac{b}{f(m+1)}\int(e+f x)^{m+1} \operatorname{Cosh}[a+b x] \operatorname{SinhIntegral}[c+d x]\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_*Sinh[a_.+b_.*x_]*SinhIntegral[c_.+d_.*x_],x_Symbol] :=
  (e+f*x)^(m+1)*Sinh[a+b*x]*SinhIntegral[c+d*x]/(f*(m+1)) -
  d/(f*(m+1))*Int[(e+f*x)^(m+1)*Sinh[a+b*x]*Sinh[c+d*x]/(c+d*x),x] -
  b/(f*(m+1))*Int[(e+f*x)^(m+1)*Cosh[a+b*x]*SinhIntegral[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && ILtQ[m,-1]
```

```
Int[(e_.+f_.*x_)^m_.*Cosh[a_.+b_.*x_]*CoshIntegral[c_.+d_.*x_],x_Symbol] :=
  (e+f*x)^(m+1)*Cosh[a+b*x]*CoshIntegral[c+d*x]/(f*(m+1)) -
  d/(f*(m+1))*Int[(e+f*x)^(m+1)*Cosh[a+b*x]*Cosh[c+d*x]/(c+d*x),x] -
  b/(f*(m+1))*Int[(e+f*x)^(m+1)*Sinh[a+b*x]*CoshIntegral[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && ILtQ[m,-1]
```

4. $\int u \operatorname{Cosh}[a+b x] \operatorname{SinhIntegral}[c+d x] \, dx$

1: $\int \operatorname{Cosh}[a+b x] \operatorname{SinhIntegral}[c+d x] \, dx$

Derivation: Integration by parts

Rule:

$$\int \operatorname{Cosh}[a+b x] \operatorname{SinhIntegral}[c+d x] \, dx \longrightarrow \frac{\operatorname{Sinh}[a+b x] \operatorname{SinhIntegral}[c+d x]}{b} - \frac{d}{b} \int \frac{\operatorname{Sinh}[a+b x] \operatorname{Sinh}[c+d x]}{c+d x} \, dx$$

Program code:

```
Int[Cosh[a_.+b_.*x_]*SinhIntegral[c_.+d_.*x_],x_Symbol] :=
  Sinh[a+b*x]*SinhIntegral[c+d*x]/b -
  d/b*Int[Sinh[a+b*x]*Sinh[c+d*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d},x]
```

```
Int[Sinh[a_.+b_.*x_]*CoshIntegral[c_.+d_.*x_],x_Symbol] :=
  Cosh[a+b*x]*CoshIntegral[c+d*x]/b -
  d/b*Int[Cosh[a+b*x]*Cosh[c+d*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d},x]
```

2. $\int (e+f x)^m \operatorname{Cosh}[a+b x] \operatorname{SinhIntegral}[c+d x] \, dx$

1: $\int (e+f x)^m \operatorname{Cosh}[a+b x] \operatorname{SinhIntegral}[c+d x] \, dx$ when $m \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $m \in \mathbb{Z}^+$, then

$$\int (e+f x)^m \operatorname{Cosh}[a+b x] \operatorname{SinhIntegral}[c+d x] \, dx \longrightarrow$$

$$\frac{(e+f x)^m \operatorname{Sinh}[a+b x] \operatorname{SinhIntegral}[c+d x]}{b}-\frac{d}{b}\int\frac{(e+f x)^m \operatorname{Sinh}[a+b x] \operatorname{Sinh}[c+d x]}{c+d x}\,dx-\frac{f m}{b}\int(e+f x)^{m-1} \operatorname{Sinh}[a+b x] \operatorname{SinhIntegral}[c+d x]\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Cosh[a_.+b_.*x_]*SinhIntegral[c_.+d_.*x_],x_Symbol] :=
  (e+f*x)^m*Sinh[a+b*x]*SinhIntegral[c+d*x]/b -
  d/b*Int[(e+f*x)^m*Sinh[a+b*x]*Sinh[c+d*x]/(c+d*x),x] -
  f*m/b*Int[(e+f*x)^(m-1)*Sinh[a+b*x]*SinhIntegral[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sinh[a_.+b_.*x_]*CoshIntegral[c_.+d_.*x_],x_Symbol] :=
  (e+f*x)^m*Cosh[a+b*x]*CoshIntegral[c+d*x]/b -
  d/b*Int[(e+f*x)^m*Cosh[a+b*x]*Cosh[c+d*x]/(c+d*x),x] -
  f*m/b*Int[(e+f*x)^(m-1)*Cosh[a+b*x]*CoshIntegral[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0]
```

2: $\int(e+f x)^m \operatorname{Cosh}[a+b x] \operatorname{SinhIntegral}[c+d x]\,dx$ when $m+1\in\mathbb{Z}^-$

Derivation: Inverted integration by parts

Rule: If $m+1\in\mathbb{Z}^-$, then

$$\int(e+f x)^m \operatorname{Cosh}[a+b x] \operatorname{SinhIntegral}[c+d x]\,dx \longrightarrow$$

$$\frac{(e+f x)^{m+1} \operatorname{Cosh}[a+b x] \operatorname{SinhIntegral}[c+d x]}{f(m+1)}-\frac{d}{f(m+1)}\int\frac{(e+f x)^{m+1} \operatorname{Cosh}[a+b x] \operatorname{Sinh}[c+d x]}{c+d x}\,dx-\frac{b}{f(m+1)}\int(e+f x)^{m+1} \operatorname{Sinh}[a+b x] \operatorname{SinhIntegral}[c+d x]\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Cosh[a_.+b_.*x_]*SinhIntegral[c_.+d_.*x_],x_Symbol] :=
  (e+f*x)^(m+1)*Cosh[a+b*x]*SinhIntegral[c+d*x]/(f*(m+1)) -
  d/(f*(m+1))*Int[(e+f*x)^(m+1)*Cosh[a+b*x]*Sinh[c+d*x]/(c+d*x),x] -
  b/(f*(m+1))*Int[(e+f*x)^(m+1)*Sinh[a+b*x]*SinhIntegral[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && ILtQ[m,-1]
```

```
Int[(e_.+f_.*x_)^m_*Sinh[a_.+b_.*x_]*CoshIntegral[c_.+d_.*x_],x_Symbol] :=
  (e+f*x)^(m+1)*Sinh[a+b*x]*CoshIntegral[c+d*x]/(f*(m+1)) -
  d/(f*(m+1))*Int[(e+f*x)^(m+1)*Sinh[a+b*x]*Cosh[c+d*x]/(c+d*x),x] -
  b/(f*(m+1))*Int[(e+f*x)^(m+1)*Cosh[a+b*x]*CoshIntegral[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && ILtQ[m,-1]
```

5. $\int u\,\text{SinhIntegral}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx$

1: $\int \text{SinhIntegral}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx$

Derivation: Integration by parts

- **Basis:** $\partial_x \text{SinhIntegral}[d\,(a+b\,\text{Log}[c\,x^n])] == \frac{b\,d\,n\,\text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]}{x\,(d\,(a+b\,\text{Log}[c\,x^n]))}$

Rule: If $m \neq -1$**, then**

$$\int \text{SinhIntegral}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx \longrightarrow x\,\text{SinhIntegral}[d\,(a+b\,\text{Log}[c\,x^n])] - b\,d\,n \int \frac{\text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]}{d\,(a+b\,\text{Log}[c\,x^n])}\,dx$$

Program code:

```
Int[SinhIntegral[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  x*SinhIntegral[d*(a+b*Log[c*x^n])] - b*d*n*Int[Sinh[d*(a+b*Log[c*x^n])]/(d*(a+b*Log[c*x^n])),x] /;
FreeQ[{a,b,c,d,n},x]
```

```
Int[CoshIntegral[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  x*CoshIntegral[d*(a+b*Log[c*x^n])] - b*d*n*Int[Cosh[d*(a+b*Log[c*x^n])]/(d*(a+b*Log[c*x^n])),x] /;
FreeQ[{a,b,c,d,n},x]
```

2: $\int \frac{\text{SinhIntegral}[d\ (a+b\ \text{Log}[c\ x^n])]}{x}\,dx$

- **Derivation: Integration by substitution**
- **Basis:** $\frac{F[\text{Log}[c\ x^n]]}{x} = \frac{1}{n}\ \text{Subst}[F[x],\ x,\ \text{Log}[c\ x^n]]\ \partial_x \text{Log}[c\ x^n]$
- **Rule:**

$$\int \frac{\text{SinhIntegral}[d\ (a+b\ \text{Log}[c\ x^n])]}{x}\,dx \to \frac{1}{n}\ \text{Subst}[\text{SinhIntegral}[d\ (a+b\ x)],\ x,\ \text{Log}[c\ x^n]]$$

- **Program code:**

```
Int[F_[d_.*(a_.+b_.*Log[c_.*x_^n_.])]/x_,x_Symbol] :=
  1/n*Subst[F[d*(a+b*x)],x,Log[c*x^n]] /;
FreeQ[{a,b,c,d,n},x] && MemberQ[{SinhIntegral,CoshIntegral},x]
```

3: $\int (e\ x)^m\ \text{SinhIntegral}[d\ (a+b\ \text{Log}[c\ x^n])]\,dx$ when $m \neq -1$

- **Derivation: Integration by parts**
- **Basis:** $\partial_x \text{SinhIntegral}[d\ (a+b\ \text{Log}[c\ x^n])] = \frac{b\ d\ n\ \text{Sinh}[d\ (a+b\ \text{Log}[c\ x^n])]}{x\ (d\ (a+b\ \text{Log}[c\ x^n]))}$
- **Rule: If** $m \neq -1$**, then**

$$\int (e\ x)^m\ \text{SinhIntegral}[d\ (a+b\ \text{Log}[c\ x^n])]\,dx \to \frac{(e\ x)^{m+1}\ \text{SinhIntegral}[d\ (a+b\ \text{Log}[c\ x^n])]}{e\ (m+1)} - \frac{b\ d\ n}{m+1} \int \frac{(e\ x)^m\ \text{Sinh}[d\ (a+b\ \text{Log}[c\ x^n])]}{d\ (a+b\ \text{Log}[c\ x^n])}\,dx$$

- **Program code:**

```
Int[(e_.*x_)^m_.*SinhIntegral[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  (e*x)^(m+1)*SinhIntegral[d*(a+b*Log[c*x^n])]/(e*(m+1)) -
  b*d*n/(m+1)*Int[(e*x)^m*Sinh[d*(a+b*Log[c*x^n])]/(d*(a+b*Log[c*x^n])),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[m,-1]
```

```
Int[(e_.*x_)^m_.*CoshIntegral[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  (e*x)^(m+1)*CoshIntegral[d*(a+b*Log[c*x^n])]/(e*(m+1)) -
  b*d*n/(m+1)*Int[(e*x)^m*Cosh[d*(a+b*Log[c*x^n])]/(d*(a+b*Log[c*x^n])),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[m,-1]
```