Rules for integrands of the form $u\,(a + b\,\mathrm{ArcSec}[c\,x])^n$

1. $\int (a + b\,\mathrm{ArcSec}[c\,x])^n\,dx$ when $n \in \mathbb{Z}^+$

1: $\int \mathrm{ArcSec}[c\,x]\,dx$

- **Reference: G&R 2.821.2, CRC 445, A&S 4.4.62**
- **Reference: G&R 2.821.1, CRC 446, A&S 4.4.61**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \mathrm{ArcSec}[c\,x]\,dx \longrightarrow x\,\mathrm{ArcSec}[c\,x] - \frac{1}{c}\int \frac{1}{x\sqrt{1-\frac{1}{c^2\,x^2}}}\,dx$$

- **Program code:**

```
Int[ArcSec[c_.*x_],x_Symbol] :=
  x*ArcSec[c*x] - 1/c*Int[1/(x*Sqrt[1-1/(c^2*x^2)]),x] /;
FreeQ[c,x]
```

```
Int[ArcCsc[c_.*x_],x_Symbol] :=
  x*ArcCsc[c*x] + 1/c*Int[1/(x*Sqrt[1-1/(c^2*x^2)]),x] /;
FreeQ[c,x]
```

2: $\int (a + b\,\text{ArcSec}[c\,x])^n\,dx$ when $n \in \mathbb{Z}^+$

- **Derivation: Integration by substitution**
- **Basis:** $1 = \frac{1}{c}\,\text{Sec}[\text{ArcSec}[c\,x]]\,\text{Tan}[\text{ArcSec}[c\,x]]\,\partial_x \text{ArcSec}[c\,x]$
- **Rule: If** $n \in \mathbb{Z}^+$**, then**

$$\int (a + b\,\text{ArcSec}[c\,x])^n\,dx \longrightarrow \frac{1}{c}\,\text{Subst}\left[\int (a + b\,x)^n\,\text{Sec}[x]\,\text{Tan}[x]\,dx,\ x,\ \text{ArcSec}[c\,x]\right]$$

- **Program code:**

```
Int[(a_.+b_.*ArcSec[c_.*x_])^n_,x_Symbol] :=
  1/c*Subst[Int[(a+b*x)^n*Sec[x]*Tan[x],x],x,ArcSec[c*x]] /;
FreeQ[{a,b,c,n},x] && IGtQ[n,0]
```

```
Int[(a_.+b_.*ArcCsc[c_.*x_])^n_,x_Symbol] :=
  -1/c*Subst[Int[(a+b*x)^n*Csc[x]*Cot[x],x],x,ArcCsc[c*x]] /;
FreeQ[{a,b,c,n},x] && IGtQ[n,0]
```

2. $\int (d\,x)^m\,(a+b\,\text{ArcSec}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

1. $\int (d\,x)^m\,(a+b\,\text{ArcSec}[c\,x])\,dx$

1: $\int \frac{a+b\,\text{ArcSec}[c\,x]}{x}\,dx$

Derivation: Integration by substitution

Basis: $\text{ArcSec}[z] = \text{ArcCos}\left[\frac{1}{z}\right]$

Basis: $\frac{F\left[\frac{1}{x}\right]}{x} = -\text{Subst}\left[\frac{F[x]}{x},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Rule:

$$\int \frac{a+b\,\text{ArcSec}[c\,x]}{x}\,dx \to \int \frac{a+b\,\text{ArcCos}\left[\frac{1}{c\,x}\right]}{x}\,dx \to -\text{Subst}\left[\int \frac{a+b\,\text{ArcCos}\left[\frac{x}{c}\right]}{x}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[(a_.+b_.*ArcSec[c_.*x_])/x_,x_Symbol] :=
  -Subst[Int[(a+b*ArcCos[x/c])/x,x],x,1/x] /;
FreeQ[{a,b,c},x]
```

```
Int[(a_.+b_.*ArcCsc[c_.*x_])/x_,x_Symbol] :=
  -Subst[Int[(a+b*ArcSin[x/c])/x,x],x,1/x] /;
FreeQ[{a,b,c},x]
```

2: $\int (d\,x)^m\,(a+b\,\text{ArcSec}[c\,x])\,dx$ when $m\neq -1$

Reference: CRC 474

Reference: CRC 477

Derivation: Integration by parts

Rule: If $m\neq -1$**, then**

$$\int (d\,x)^m\,(a+b\,\text{ArcSec}[c\,x])\,dx \longrightarrow \frac{(d\,x)^{m+1}\,(a+b\,\text{ArcSec}[c\,x])}{d\,(m+1)} - \frac{b\,d}{c\,(m+1)}\int \frac{(d\,x)^{m-1}}{\sqrt{1-\frac{1}{c^2\,x^2}}}\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcSec[c_.*x_]),x_Symbol] :=
  (d*x)^(m+1)*(a+b*ArcSec[c*x])/(d*(m+1)) -
  b*d/(c*(m+1))*Int[(d*x)^(m-1)/Sqrt[1-1/(c^2*x^2)],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1]
```

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcCsc[c_.*x_]),x_Symbol] :=
  (d*x)^(m+1)*(a+b*ArcCsc[c*x])/(d*(m+1)) +
  b*d/(c*(m+1))*Int[(d*x)^(m-1)/Sqrt[1-1/(c^2*x^2)],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1]
```

2: $\int x^m\,(a+b\,\text{ArcSec}[c\,x])^n\,dx$ when $n \in \mathbb{Z} \wedge m \in \mathbb{Z} \wedge (n>0 \vee m<-1)$

Derivation: Integration by substitution

Basis: If $m \in \mathbb{Z}$, **then** $x^m\,F[\text{ArcSec}[c\,x]] = \frac{1}{c^{m+1}}\,\text{Subst}\left[F[x]\,\text{Sec}[x]^{m+1}\,\text{Tan}[x],\ x,\ \text{ArcSec}[c\,x]\right]\,\partial_x \text{ArcSec}[c\,x]$

Rule: If $n \in \mathbb{Z} \wedge m \in \mathbb{Z} \wedge (n>0 \vee m<-1)$, **then**

$$\int x^m\,(a+b\,\text{ArcSec}[c\,x])^n\,dx \longrightarrow \frac{1}{c^{m+1}}\,\text{Subst}\left[\int (a+b\,x)^n\,\text{Sec}[x]^{m+1}\,\text{Tan}[x]\,dx,\ x,\ \text{ArcSec}[c\,x]\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcSec[c_.*x_])^n_,x_Symbol] :=
  1/c^(m+1)*Subst[Int[(a+b*x)^n*Sec[x]^(m+1)*Tan[x],x],x,ArcSec[c*x]] /;
FreeQ[{a,b,c},x] && IntegerQ[n] && IntegerQ[m] && (GtQ[n,0] || LtQ[m,-1])
```

```
Int[x_^m_.*(a_.+b_.*ArcCsc[c_.*x_])^n_,x_Symbol] :=
  -1/c^(m+1)*Subst[Int[(a+b*x)^n*Csc[x]^(m+1)*Cot[x],x],x,ArcCsc[c*x]] /;
FreeQ[{a,b,c},x] && IntegerQ[n] && IntegerQ[m] && (GtQ[n,0] || LtQ[m,-1])
```

3. $\int (d+e\,x)^m\,(a+b\,\text{ArcSec}[c\,x])\,dx$

1: $\int \frac{a+b\,\text{ArcSec}[c\,x]}{d+e\,x}\,dx$

Derivation: Integration by parts

- **Basis:** $\frac{1}{d+e\,x} = \frac{1}{e}\,\partial_x\left(\text{Log}\left[1+\frac{\left(e-\sqrt{-c^2\,d^2+e^2}\right)\,e^{i\,\text{ArcSec}[c\,x]}}{c\,d}\right]+\text{Log}\left[1+\frac{\left(e+\sqrt{-c^2\,d^2+e^2}\right)\,e^{i\,\text{ArcSec}[c\,x]}}{c\,d}\right]-\text{Log}\left[1+e^{2\,i\,\text{ArcSec}[c\,x]}\right]\right)$
- **Basis:** $\frac{1}{d+e\,x} = \frac{1}{e}\,\partial_x\left(\text{Log}\left[1-\frac{i\left(e-\sqrt{-c^2\,d^2+e^2}\right)\,e^{i\,\text{ArcCsc}[c\,x]}}{c\,d}\right]+\text{Log}\left[1-\frac{i\left(e+\sqrt{-c^2\,d^2+e^2}\right)\,e^{i\,\text{ArcCsc}[c\,x]}}{c\,d}\right]-\text{Log}\left[1-e^{2\,i\,\text{ArcCsc}[c\,x]}\right]\right)$

Rule:

$$\int \frac{a+b\,\text{ArcSec}[c\,x]}{d+e\,x}\,dx \longrightarrow$$

$$\frac{(a+b\,\text{ArcSec}[c\,x])\,\text{Log}\left[1+\frac{\left(e-\sqrt{-c^2\,d^2+e^2}\right)\,e^{i\,\text{ArcSec}[c\,x]}}{c\,d}\right]}{e}+\frac{(a+b\,\text{ArcSec}[c\,x])\,\text{Log}\left[1+\frac{\left(e+\sqrt{-c^2\,d^2+e^2}\right)\,e^{i\,\text{ArcSec}[c\,x]}}{c\,d}\right]}{e}-$$

$$\frac{(a+b\,\text{ArcSec}[c\,x])\,\text{Log}\left[1+e^{2\,i\,\text{ArcSec}[c\,x]}\right]}{e}-\frac{b}{c\,e}\int\frac{\text{Log}\left[1+\frac{\left(e-\sqrt{-c^2\,d^2+e^2}\right)\,e^{i\,\text{ArcSec}[c\,x]}}{c\,d}\right]}{x^2\,\sqrt{1-\frac{1}{c^2\,x^2}}}\,dx-$$

$$\frac{b}{c\,e}\int\frac{\text{Log}\left[1+\frac{\left(e+\sqrt{-c^2\,d^2+e^2}\right)\,e^{i\,\text{ArcSec}[c\,x]}}{c\,d}\right]}{x^2\,\sqrt{1-\frac{1}{c^2\,x^2}}}\,dx+\frac{b}{c\,e}\int\frac{\text{Log}\left[1+e^{2\,i\,\text{ArcSec}[c\,x]}\right]}{x^2\,\sqrt{1-\frac{1}{c^2\,x^2}}}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcSec[c_.*x_])/(d_.+e_.*x_),x_Symbol] :=
  (a+b*ArcSec[c*x])*Log[1+(e-Sqrt[-c^2*d^2+e^2])*E^(I*ArcSec[c*x])/(c*d)]/e +
  (a+b*ArcSec[c*x])*Log[1+(e+Sqrt[-c^2*d^2+e^2])*E^(I*ArcSec[c*x])/(c*d)]/e -
  (a+b*ArcSec[c*x])*Log[1+E^(2*I*ArcSec[c*x])]/e -
  b/(c*e)*Int[Log[1+(e-Sqrt[-c^2*d^2+e^2])*E^(I*ArcSec[c*x])/(c*d)]/(x^2*Sqrt[1-1/(c^2*x^2)]),x] -
  b/(c*e)*Int[Log[1+(e+Sqrt[-c^2*d^2+e^2])*E^(I*ArcSec[c*x])/(c*d)]/(x^2*Sqrt[1-1/(c^2*x^2)]),x] +
  b/(c*e)*Int[Log[1+E^(2*I*ArcSec[c*x])]/(x^2*Sqrt[1-1/(c^2*x^2)]),x] /;
FreeQ[{a,b,c,d,e},x]
```

```
Int[(a_.+b_.*ArcCsc[c_.*x_])/(d_.+e_.*x_),x_Symbol] :=
  (a+b*ArcCsc[c*x])*Log[1-I*(e-Sqrt[-c^2*d^2+e^2])*E^(I*ArcCsc[c*x])/(c*d)]/e +
  (a+b*ArcCsc[c*x])*Log[1-I*(e+Sqrt[-c^2*d^2+e^2])*E^(I*ArcCsc[c*x])/(c*d)]/e -
  (a+b*ArcCsc[c*x])*Log[1-E^(2*I*ArcCsc[c*x])]/e +
  b/(c*e)*Int[Log[1-I*(e-Sqrt[-c^2*d^2+e^2])*E^(I*ArcCsc[c*x])/(c*d)]/(x^2*Sqrt[1-1/(c^2*x^2)]),x] +
  b/(c*e)*Int[Log[1-I*(e+Sqrt[-c^2*d^2+e^2])*E^(I*ArcCsc[c*x])/(c*d)]/(x^2*Sqrt[1-1/(c^2*x^2)]),x] -
  b/(c*e)*Int[Log[1-E^(2*I*ArcCsc[c*x])]/(x^2*Sqrt[1-1/(c^2*x^2)]),x] /;
FreeQ[{a,b,c,d,e},x]
```

2: $\int (d+e\,x)^m\,(a+b\,\text{ArcSec}[c\,x])\,dx$ when $m \neq -1$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{ArcSec}[c\,x]) = \frac{b}{c\,x^2\sqrt{1-\frac{1}{c^2 x^2}}}$

Rule: If $m \neq -1$**, then**

$$\int (d+e\,x)^m\,(a+b\,\text{ArcSec}[c\,x])\,dx \longrightarrow \frac{(d+e\,x)^{m+1}\,(a+b\,\text{ArcSec}[c\,x])}{e\,(m+1)} - \frac{b}{c\,e\,(m+1)}\int \frac{(d+e\,x)^{m+1}}{x^2\sqrt{1-\frac{1}{c^2 x^2}}}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcSec[c_.*x_]),x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*ArcSec[c*x])/(e*(m+1)) -
  b/(c*e*(m+1))*Int[(d+e*x)^(m+1)/(x^2*Sqrt[1-1/(c^2*x^2)]),x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[m,-1]
```

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcCsc[c_.*x_]),x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*ArcCsc[c*x])/(e*(m+1)) +
  b/(c*e*(m+1))*Int[(d+e*x)^(m+1)/(x^2*Sqrt[1-1/(c^2*x^2)]),x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[m,-1]
```

4. $\int (d+e x^2)^p (a+b \operatorname{ArcSec}[c x])^n \, dx$ when $n \in \mathbb{Z}^+$

1: $\int (d+e x^2)^p (a+b \operatorname{ArcSec}[c x]) \, dx$ when $p \in \mathbb{Z}^+ \bigvee p+\frac{1}{2} \in \mathbb{Z}^-$

Derivation: Integration by parts and piecewise constant extraction

- **Basis:** $\partial_x (a+b \operatorname{ArcSec}[c x]) == \frac{b c}{\sqrt{c^2 x^2} \sqrt{c^2 x^2-1}}$

  Basis: $\partial_x \frac{x}{\sqrt{c^2 x^2}} == 0$

- **Note: If** $p \in \mathbb{Z}^+ \bigvee p+\frac{1}{2} \in \mathbb{Z}^-$, **then** $\int (d+e x^2)^p \, dx$ **is expressible as an algebraic function not involving logarithms, inverse trig or inverse hyperbolic functions.**

- **Rule: If** $p \in \mathbb{Z}^+ \bigvee p+\frac{1}{2} \in \mathbb{Z}^-$, **let** $u = \int (d+e x^2)^p \, dx$, **then**

$$\int (d+e x^2)^p (a+b \operatorname{ArcSec}[c x]) \, dx \to u (a+b \operatorname{ArcSec}[c x]) - b c \int \frac{u}{\sqrt{c^2 x^2} \sqrt{c^2 x^2-1}} \, dx \to u (a+b \operatorname{ArcSec}[u]) - \frac{b c x}{\sqrt{c^2 x^2}} \int \frac{u}{x \sqrt{c^2 x^2-1}} \, dx$$

- **Program code:**

```
Int[(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcSec[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[(a+b*ArcSec[c*x]),u,x] - b*c*x/Sqrt[c^2*x^2]*Int[SimplifyIntegrand[u/(x*Sqrt[c^2*x^2-1]),x],x]] /;
FreeQ[{a,b,c,d,e},x] && (IGtQ[p,0] || ILtQ[p+1/2,0])
```

```
Int[(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcCsc[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[(a+b*ArcCsc[c*x]),u,x] + b*c*x/Sqrt[c^2*x^2]*Int[SimplifyIntegrand[u/(x*Sqrt[c^2*x^2-1]),x],x]] /;
FreeQ[{a,b,c,d,e},x] && (IGtQ[p,0] || ILtQ[p+1/2,0])
```

2: $\int (d+e x^2)^p (a+b \operatorname{ArcSec}[c x])^n \, dx$ when $n \in \mathbb{Z}^+ \bigwedge p \in \mathbb{Z}$

Derivation: Integration by substitution

- **Basis:** $\operatorname{ArcSec}[z] == \operatorname{ArcCos}\left[\frac{1}{z}\right]$

- **Basis:** $F\left[\frac{1}{x}\right] == -\operatorname{Subst}\left[\frac{F[x]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

- **Rule: If** $n \in \mathbb{Z}^+ \bigwedge p \in \mathbb{Z}$, **then**

$$\int \left(d+e\,x^2\right)^p \left(a+b\,\text{ArcSec}[c\,x]\right)^n dx \longrightarrow \int \left(\frac{1}{x}\right)^{-2\,p} \left(e+\frac{d}{x^2}\right)^p \left(a+b\,\text{ArcCos}\left[\frac{1}{c\,x}\right]\right)^n dx$$

$$\longrightarrow -\text{Subst}\left[\int \frac{\left(e+d\,x^2\right)^p \left(a+b\,\text{ArcCos}\left[\frac{x}{c}\right]\right)^n}{x^{2\,(p+1)}}\,dx,\ x,\ \frac{1}{x}\right]$$

– **Program code:**

```
Int[(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcSec[c_.*x_])^n_.,x_Symbol] :=
  -Subst[Int[(e+d*x^2)^p*(a+b*ArcCos[x/c])^n/x^(2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && IntegerQ[p]
```

```
Int[(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcCsc[c_.*x_])^n_.,x_Symbol] :=
  -Subst[Int[(e+d*x^2)^p*(a+b*ArcSin[x/c])^n/x^(2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && IntegerQ[p]
```

3. $\int (d+e x^2)^p (a+b\,\mathrm{ArcSec}[c x])^n \, dx$ when $n \in \mathbb{Z}^+ \bigwedge c^2 d+e == 0 \bigwedge p+\frac{1}{2} \in \mathbb{Z}$

1: $\int (d+e x^2)^p (a+b\,\mathrm{ArcSec}[c x])^n \, dx$ when $n \in \mathbb{Z}^+ \bigwedge c^2 d+e == 0 \bigwedge p+\frac{1}{2} \in \mathbb{Z} \bigwedge e>0 \bigwedge d<0$

Derivation: Piecewise constant extraction and integration by substitution

- Basis: $\partial_x \frac{\sqrt{d+e x^2}}{x \sqrt{e+\frac{d}{x^2}}} == 0$
- Basis: $\mathrm{ArcSec}[z] == \mathrm{ArcCos}\left[\frac{1}{z}\right]$
- Basis: $F\left[\frac{1}{x}\right] == -\mathrm{Subst}\left[\frac{F[x]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$
- Basis: If $e>0 \land d<0$, then $\frac{\sqrt{d+e x^2}}{\sqrt{e+\frac{d}{x^2}}} == \sqrt{x^2}$
- Rule: If $n \in \mathbb{Z}^+ \bigwedge c^2 d+e == 0 \bigwedge p+\frac{1}{2} \in \mathbb{Z} \bigwedge e>0 \bigwedge d<0$, then

$$\int (d+e x^2)^p (a+b\,\mathrm{ArcSec}[c x])^n \, dx \longrightarrow \frac{\sqrt{d+e x^2}}{x \sqrt{e+\frac{d}{x^2}}} \int \left(\frac{1}{x}\right)^{-2 p} \left(e+\frac{d}{x^2}\right)^p \left(a+b\,\mathrm{ArcCos}\left[\frac{1}{c x}\right]\right)^n dx$$

$$\longrightarrow -\frac{\sqrt{x^2}}{x} \mathrm{Subst}\left[\int \frac{(e+d x^2)^p \left(a+b\,\mathrm{ArcCos}\left[\frac{x}{c}\right]\right)^n}{x^{2 (p+1)}} dx, x, \frac{1}{x}\right]$$

Program code:

```
Int[(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcSec[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[x^2]/x*Subst[Int[(e+d*x^2)^p*(a+b*ArcCos[x/c])^n/x^(2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[p+1/2] && GtQ[e,0] && LtQ[d,0]
```

```
Int[(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcCsc[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[x^2]/x*Subst[Int[(e+d*x^2)^p*(a+b*ArcSin[x/c])^n/x^(2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[p+1/2] && GtQ[e,0] && LtQ[d,0]
```

2: $\int (d+e x^2)^p (a+b \operatorname{ArcSec}[c x])^n \, dx$ when $n \in \mathbb{Z}^+ \bigwedge c^2 d+e == 0 \bigwedge p+\frac{1}{2} \in \mathbb{Z} \bigwedge \neg (e>0 \land d<0)$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x \frac{\sqrt{d+e x^2}}{x \sqrt{e+\frac{d}{x^2}}} == 0$
- **Basis:** $\operatorname{ArcSec}[z] == \operatorname{ArcCos}\left[\frac{1}{z}\right]$
- **Basis:** $F\left[\frac{1}{x}\right] == -\operatorname{Subst}\left[\frac{F[x]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$
- **Rule: If** $n \in \mathbb{Z}^+ \bigwedge c^2 d+e == 0 \bigwedge p+\frac{1}{2} \in \mathbb{Z} \bigwedge \neg (e>0 \land d<0)$**, then**

$$\int (d+e x^2)^p (a+b \operatorname{ArcSec}[c x])^n \, dx \to \frac{\sqrt{d+e x^2}}{x \sqrt{e+\frac{d}{x^2}}} \int \left(\frac{1}{x}\right)^{-2p} \left(e+\frac{d}{x^2}\right)^p \left(a+b \operatorname{ArcCos}\left[\frac{1}{c x}\right]\right)^n dx$$

$$\to -\frac{\sqrt{d+e x^2}}{x \sqrt{e+\frac{d}{x^2}}} \operatorname{Subst}\left[\int \frac{(e+d x^2)^p \left(a+b \operatorname{ArcCos}\left[\frac{x}{c}\right]\right)^n}{x^{2(p+1)}} dx, x, \frac{1}{x}\right]$$

Program code:

```
Int[(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcSec[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[d+e*x^2]/(x*Sqrt[e+d/x^2])*Subst[Int[(e+d*x^2)^p*(a+b*ArcCos[x/c])^n/x^(2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[p+1/2] && Not[GtQ[e,0] && LtQ[d,0]]
```

```
Int[(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcCsc[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[d+e*x^2]/(x*Sqrt[e+d/x^2])*Subst[Int[(e+d*x^2)^p*(a+b*ArcSin[x/c])^n/x^(2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[p+1/2] && Not[GtQ[e,0] && LtQ[d,0]]
```

5. $\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSec}[c\,x])^n\,dx$ when $n \in \mathbb{Z}^+$

1. $\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSec}[c\,x])\,dx$ when

$$\left(p \in \mathbb{Z}^+ \bigwedge \neg \left(\frac{m-1}{2} \in \mathbb{Z}^- \bigwedge m+2p+3>0\right)\right) \bigvee \left(\frac{m+1}{2} \in \mathbb{Z}^+ \bigwedge \neg (p \in \mathbb{Z}^- \wedge m+2p+3>0)\right) \bigvee \left(\frac{m+2p+1}{2} \in \mathbb{Z}^- \bigwedge \frac{m-1}{2} \notin \mathbb{Z}^-\right)$$

1: $\int x \left(d+e\,x^2\right)^p (a+b\,\text{ArcSec}[c\,x])\,dx$ when $p \neq -1$

Derivation: Integration by parts and piecewise constant extraction

- **Basis:** $x\left(d+e\,x^2\right)^p = \partial_x \frac{\left(d+e\,x^2\right)^{p+1}}{2\,e\,(p+1)}$
- **Basis:** $\partial_x (a+b\,\text{ArcSec}[c\,x]) = \frac{b\,c}{\sqrt{c^2\,x^2}\,\sqrt{c^2\,x^2-1}}$
- **Basis:** $\partial_x \frac{x}{\sqrt{c^2\,x^2}} = 0$
- **Rule:** If $p \neq -1$, then

$$\int x\left(d+e\,x^2\right)^p (a+b\,\text{ArcSec}[c\,x])\,dx \longrightarrow \frac{\left(d+e\,x^2\right)^{p+1} (a+b\,\text{ArcSec}[c\,x])}{2\,e\,(p+1)} - \frac{b\,c}{2\,e\,(p+1)} \int \frac{\left(d+e\,x^2\right)^{p+1}}{\sqrt{c^2\,x^2}\,\sqrt{c^2\,x^2-1}}\,dx$$

$$\longrightarrow \frac{\left(d+e\,x^2\right)^{p+1} (a+b\,\text{ArcSec}[c\,x])}{2\,e\,(p+1)} - \frac{b\,c\,x}{2\,e\,(p+1)\,\sqrt{c^2\,x^2}} \int \frac{\left(d+e\,x^2\right)^{p+1}}{x\,\sqrt{c^2\,x^2-1}}\,dx$$

Program code:

```
Int[x_*(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcSec[c_.*x_]),x_Symbol] :=
  (d+e*x^2)^(p+1)*(a+b*ArcSec[c*x])/(2*e*(p+1)) -
  b*c*x/(2*e*(p+1)*Sqrt[c^2*x^2])*Int[(d+e*x^2)^(p+1)/(x*Sqrt[c^2*x^2-1]),x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[p,-1]
```

```
Int[x_*(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcCsc[c_.*x_]),x_Symbol] :=
  (d+e*x^2)^(p+1)*(a+b*ArcCsc[c*x])/(2*e*(p+1)) +
  b*c*x/(2*e*(p+1)*Sqrt[c^2*x^2])*Int[(d+e*x^2)^(p+1)/(x*Sqrt[c^2*x^2-1]),x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[p,-1]
```

2: $\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSec}[c\,x])\,dx$ when

$$\left(p\in\mathbb{Z}^+ \bigwedge \neg\left(\frac{m-1}{2}\in\mathbb{Z}^- \bigwedge m+2\,p+3>0\right)\right) \bigvee \left(\frac{m+1}{2}\in\mathbb{Z}^+ \bigwedge \neg\left(p\in\mathbb{Z}^- \wedge m+2\,p+3>0\right)\right) \bigvee \left(\frac{m+2\,p+1}{2}\in\mathbb{Z}^- \bigwedge \frac{m-1}{2}\notin\mathbb{Z}^-\right)$$

Derivation: Integration by parts and piecewise constant extraction

- **Basis:** $\partial_x (a+b\,\text{ArcSec}[c\,x]) == \frac{b\,c}{\sqrt{c^2\,x^2}\,\sqrt{c^2\,x^2-1}}$

  Basis: $\partial_x \frac{x}{\sqrt{c^2\,x^2}} == 0$

- **Note: If** $\left(p\in\mathbb{Z}^+ \bigwedge \neg\left(\frac{m-1}{2}\in\mathbb{Z}^- \bigwedge m+2\,p+3>0\right)\right) \bigvee \left(\frac{m+1}{2}\in\mathbb{Z}^+ \bigwedge \neg\left(p\in\mathbb{Z}^- \wedge m+2\,p+3>0\right)\right) \bigvee \left(\frac{m+2\,p+1}{2}\in\mathbb{Z}^- \bigwedge \frac{m-1}{2}\notin\mathbb{Z}^-\right)$,
  then $\int (f\,x)^m \left(d+e\,x^2\right)^p dx$ **is expressible as an algebraic function not involving logarithms, inverse trig or inverse hyperbolic functions.**

- **Rule: If** $\left(p\in\mathbb{Z}^+ \bigwedge \neg\left(\frac{m-1}{2}\in\mathbb{Z}^- \bigwedge m+2\,p+3>0\right)\right) \bigvee \left(\frac{m+1}{2}\in\mathbb{Z}^+ \bigwedge \neg\left(p\in\mathbb{Z}^- \wedge m+2\,p+3>0\right)\right) \bigvee \left(\frac{m+2\,p+1}{2}\in\mathbb{Z}^- \bigwedge \frac{m-1}{2}\notin\mathbb{Z}^-\right)$,
  let $u = \int (f\,x)^m \left(d+e\,x^2\right)^p dx$**, then**

$$\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSec}[c\,x])\,dx \longrightarrow u\,(a+b\,\text{ArcSec}[c\,x]) - b\,c\int \frac{u}{\sqrt{c^2\,x^2}\,\sqrt{c^2\,x^2-1}}\,dx$$

$$\longrightarrow u\,(a+b\,\text{ArcSec}[u]) - \frac{b\,c\,x}{\sqrt{c^2\,x^2}}\int \frac{u}{x\,\sqrt{c^2\,x^2-1}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcSec[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^2)^p,x]},
  Dist[(a+b*ArcSec[c*x]),u,x] - b*c*x/Sqrt[c^2*x^2]*Int[SimplifyIntegrand[u/(x*Sqrt[c^2*x^2-1]),x],x]] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && (
  IGtQ[p,0] && Not[ILtQ[(m-1)/2,0] && GtQ[m+2*p+3,0]] ||
  IGtQ[(m+1)/2,0] && Not[ILtQ[p,0] && GtQ[m+2*p+3,0]] ||
  ILtQ[(m+2*p+1)/2,0] && Not[ILtQ[(m-1)/2,0]])
```

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcCsc[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^2)^p,x]},
  Dist[(a+b*ArcCsc[c*x]),u,x] + b*c*x/Sqrt[c^2*x^2]*Int[SimplifyIntegrand[u/(x*Sqrt[c^2*x^2-1]),x],x]] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && (
  IGtQ[p,0] && Not[ILtQ[(m-1)/2,0] && GtQ[m+2*p+3,0]] ||
  IGtQ[(m+1)/2,0] && Not[ILtQ[p,0] && GtQ[m+2*p+3,0]] ||
  ILtQ[(m+2*p+1)/2,0] && Not[ILtQ[(m-1)/2,0]])
```

2: $\int x^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSec}[c\,x])^n \, dx$ when $n \in \mathbb{Z}^+ \wedge m \in \mathbb{Z} \wedge p \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{ArcSec}[z] = \text{ArcCos}\left[\frac{1}{z}\right]$

Basis: $F\left[\frac{1}{x}\right] = -\text{Subst}\left[\frac{F[x]}{x^2},\, x,\, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Rule: If $n \in \mathbb{Z}^+ \wedge m \in \mathbb{Z} \wedge p \in \mathbb{Z}$, then

$$\int x^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSec}[c\,x])^n \, dx \to \int \left(\frac{1}{x}\right)^{-m-2p} \left(e+\frac{d}{x^2}\right)^p \left(a+b\,\text{ArcCos}\left[\frac{1}{c\,x}\right]\right)^n dx$$

$$\to -\text{Subst}\left[\int \frac{\left(e+d\,x^2\right)^p \left(a+b\,\text{ArcCos}\left[\frac{x}{c}\right]\right)^n}{x^{m+2\,(p+1)}} \, dx,\, x,\, \frac{1}{x}\right]$$

Program code:

```
Int[x_^m_.*(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcSec[c_.*x_])^n_.,x_Symbol] :=
  -Subst[Int[(e+d*x^2)^p*(a+b*ArcCos[x/c])^n/x^(m+2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && IntegerQ[m] && IntegerQ[p]
```

```
Int[x_^m_.*(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcCsc[c_.*x_])^n_.,x_Symbol] :=
  -Subst[Int[(e+d*x^2)^p*(a+b*ArcSin[x/c])^n/x^(m+2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && IntegerQ[m] && IntegerQ[p]
```

3. $\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSec}[c x])^n \, dx$ when $n \in \mathbb{Z}^+ \bigwedge c^2 d+e==0 \bigwedge m \in \mathbb{Z} \bigwedge p+\frac{1}{2} \in \mathbb{Z}$

1: $\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSec}[c x])^n \, dx$ when $n \in \mathbb{Z}^+ \bigwedge c^2 d+e==0 \bigwedge m \in \mathbb{Z} \bigwedge p+\frac{1}{2} \in \mathbb{Z} \bigwedge e>0 \bigwedge d<0$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x \frac{\sqrt{d+e x^2}}{x \sqrt{e+\frac{d}{x^2}}}==0$
- **Basis:** $\operatorname{ArcSec}[z]==\operatorname{ArcCos}\left[\frac{1}{z}\right]$
- **Basis:** $F\left[\frac{1}{x}\right]==-\operatorname{Subst}\left[\frac{F[x]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$
- **Basis: If** $e>0 \land d<0$**, then** $\frac{\sqrt{d+e x^2}}{\sqrt{e+\frac{d}{x^2}}}==\sqrt{x^2}$
- **Rule: If** $n \in \mathbb{Z}^+ \bigwedge c^2 d+e==0 \bigwedge m \in \mathbb{Z} \bigwedge p+\frac{1}{2} \in \mathbb{Z} \bigwedge e>0 \bigwedge d<0$**, then**

$$\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSec}[c x])^n \, dx \longrightarrow \frac{\sqrt{d+e x^2}}{x \sqrt{e+\frac{d}{x^2}}} \int \left(\frac{1}{x}\right)^{-m-2 p} \left(e+\frac{d}{x^2}\right)^p \left(a+b \operatorname{ArcCos}\left[\frac{1}{c x}\right]\right)^n dx$$

$$\longrightarrow -\frac{\sqrt{x^2}}{x} \operatorname{Subst}\left[\int \frac{\left(e+d x^2\right)^p \left(a+b \operatorname{ArcCos}\left[\frac{x}{c}\right]\right)^n}{x^{m+2 (p+1)}} dx, x, \frac{1}{x}\right]$$

Program code:

```
Int[x_^m_.*(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcSec[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[x^2]/x*Subst[Int[(e+d*x^2)^p*(a+b*ArcCos[x/c])^n/x^(m+2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[m] && IntegerQ[p+1/2] && GtQ[e,0] && LtQ[d,0]
```

```
Int[x_^m_.*(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcCsc[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[x^2]/x*Subst[Int[(e+d*x^2)^p*(a+b*ArcSin[x/c])^n/x^(m+2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[m] && IntegerQ[p+1/2] && GtQ[e,0] && LtQ[d,0]
```

2: $\int x^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSec}[c\,x])^n \,dx$ when $n\in\mathbb{Z}^+ \bigwedge c^2 d+e==0 \bigwedge m\in\mathbb{Z} \bigwedge p+\frac{1}{2}\in\mathbb{Z} \bigwedge \neg\,(e>0 \land d<0)$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x \frac{\sqrt{d+e\,x^2}}{x\sqrt{e+\frac{d}{x^2}}} == 0$
- **Basis:** $\text{ArcSec}[z] == \text{ArcCos}\left[\frac{1}{z}\right]$
- **Basis:** $F\left[\frac{1}{x}\right] == -\text{Subst}\left[\frac{F[x]}{x^2},\, x,\, \frac{1}{x}\right] \partial_x \frac{1}{x}$
- **Rule:** If $n\in\mathbb{Z}^+ \bigwedge c^2 d+e==0 \bigwedge m\in\mathbb{Z} \bigwedge p+\frac{1}{2}\in\mathbb{Z} \bigwedge \neg\,(e>0 \land d<0)$, then

$$\int x^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSec}[c\,x])^n \,dx \to \frac{\sqrt{d+e\,x^2}}{x\sqrt{e+\frac{d}{x^2}}} \int \left(\frac{1}{x}\right)^{-m-2p} \left(e+\frac{d}{x^2}\right)^p \left(a+b\,\text{ArcCos}\left[\frac{1}{c\,x}\right]\right)^n dx$$

$$\to -\frac{\sqrt{d+e\,x^2}}{x\sqrt{e+\frac{d}{x^2}}}\,\text{Subst}\left[\int \frac{\left(e+d\,x^2\right)^p \left(a+b\,\text{ArcCos}\left[\frac{x}{c}\right]\right)^n}{x^{m+2(p+1)}}\,dx,\, x,\, \frac{1}{x}\right]$$

- **Program code:**

```
Int[x_^m_.*(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcSec[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[d+e*x^2]/(x*Sqrt[e+d/x^2])*Subst[Int[(e+d*x^2)^p*(a+b*ArcCos[x/c])^n/x^(m+2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[m] && IntegerQ[p+1/2] && Not[GtQ[e,0] && LtQ[d,0]]
```

```
Int[x_^m_.*(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcCsc[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[d+e*x^2]/(x*Sqrt[e+d/x^2])*Subst[Int[(e+d*x^2)^p*(a+b*ArcSin[x/c])^n/x^(m+2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[m] && IntegerQ[p+1/2] && Not[GtQ[e,0] && LtQ[d,0]]
```

6: $\int u\,(a+b\,\text{ArcSec}[c\,x])\,dx$ when $\int u\,dx$ is free of inverse functions

- **Derivation: Integration by parts**
- **Basis:** $\partial_x (a+b\,\text{ArcSec}[c\,x]) == \frac{b}{c\,x^2\sqrt{1-\frac{1}{c^2 x^2}}}$
- **Rule:** Let $v \to \int u\,dx$, if v is free of inverse functions, then

$$\int u\,(a+b\,\text{ArcSec}[c\,x])\,dx \longrightarrow v\,(a+b\,\text{ArcSec}[c\,x]) - \frac{b}{c}\int \frac{v}{x^2\sqrt{1-\frac{1}{c^2\,x^2}}}\,dx$$

- **Program code:**

```
Int[u_*(a_.+b_.*ArcSec[c_.*x_]),x_Symbol] :=
  With[{v=IntHide[u,x]},
  Dist[(a+b*ArcSec[c*x]),v,x] -
  b/c*Int[SimplifyIntegrand[v/(x^2*Sqrt[1-1/(c^2*x^2)]),x],x] /;
 InverseFunctionFreeQ[v,x]] /;
FreeQ[{a,b,c},x]
```

```
Int[u_*(a_.+b_.*ArcCsc[c_.*x_]),x_Symbol] :=
  With[{v=IntHide[u,x]},
  Dist[(a+b*ArcCsc[c*x]),v,x] +
  b/c*Int[SimplifyIntegrand[v/(x^2*Sqrt[1-1/(c^2*x^2)]),x],x] /;
 InverseFunctionFreeQ[v,x]] /;
FreeQ[{a,b,c},x]
```

X: $\int u\,(a+b\,\text{ArcSec}[c\,x])^n\,dx$

- **Rule:**

$$\int u\,(a+b\,\text{ArcSec}[c\,x])^n\,dx \longrightarrow \int u\,(a+b\,\text{ArcSec}[c\,x])^n\,dx$$

- **Program code:**

```
Int[u_.*(a_.+b_.*ArcSec[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[u*(a+b*ArcSec[c*x])^n,x] /;
FreeQ[{a,b,c,n},x]
```

```
Int[u_.*(a_.+b_.*ArcCsc[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[u*(a+b*ArcCsc[c*x])^n,x] /;
FreeQ[{a,b,c,n},x]
```