Rules for integrands of the form $(g \operatorname{Tan}[e+f x])^p (a+b \operatorname{Tan}[e+f x])^m (c+d \operatorname{Tan}[e+f x])^n$

X: $\int (g \operatorname{Tan}[e+f x])^p (a+b \operatorname{Tan}[e+f x])^m (c+d \operatorname{Tan}[e+f x])^n \, dx$

- **Rule:**

$$\int (g \operatorname{Tan}[e+f x])^p (a+b \operatorname{Tan}[e+f x])^m (c+d \operatorname{Tan}[e+f x])^n \, dx \longrightarrow \int (g \operatorname{Tan}[e+f x])^p (a+b \operatorname{Tan}[e+f x])^m (c+d \operatorname{Tan}[e+f x])^n \, dx$$

- **Program code:**

```
Int[(g_.*tan[e_.+f_.*x_])^p_.*(a_+b_.*tan[e_.+f_.*x_])^m_*(c_+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  Unintegrable[(g*Tan[e+f*x])^p*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x]
```

Rules for integrands of the form $(g \operatorname{Tan}[e+f x]^q)^p (a+b \operatorname{Tan}[e+f x])^m (c+d \operatorname{Tan}[e+f x])^n$

1: $\int (g \operatorname{Cot}[e+f x])^p (a+b \operatorname{Tan}[e+f x])^m (c+d \operatorname{Tan}[e+f x])^n \, dx$ when $p \notin \mathbb{Z} \land m \in \mathbb{Z} \land n \in \mathbb{Z}$

- **Derivation: Algebraic normalization**

- **Basis: If** $m \in \mathbb{Z} \land n \in \mathbb{Z}$**, then** $(a+b \operatorname{Tan}[z])^m (c+d \operatorname{Tan}[z])^n == \frac{g^{m+n} (b+a \operatorname{Cot}[z])^m (d+c \operatorname{Cot}[z])^n}{(g \operatorname{Cot}[z])^{m+n}}$

- **Rule: If** $p \notin \mathbb{Z} \land m \in \mathbb{Z} \land n \in \mathbb{Z}$**, then**

$$\int (g \operatorname{Cot}[e+f x])^p (a+b \operatorname{Tan}[e+f x])^m (c+d \operatorname{Tan}[e+f x])^n \, dx \longrightarrow g^{m+n} \int (g \operatorname{Cot}[e+f x])^{p-m-n} (b+a \operatorname{Cot}[e+f x])^m (d+c \operatorname{Cot}[e+f x])^n \, dx$$

- **Program code:**

```
Int[(g_.*cot[e_.+f_.*x_])^p_*(a_.+b_.*tan[e_.+f_.*x_])^m_.*(c_+d_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  g^(m+n)*Int[(g*Cot[e+f*x])^(p-m-n)*(b+a*Cot[e+f*x])^m*(d+c*Cot[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && Not[IntegerQ[p]] && IntegerQ[m] && IntegerQ[n]
```

```
Int[(g_.*tan[e_.+f_.*x_])^p_*(a_.+b_.*cot[e_.+f_.*x_])^m_.*(c_+d_.*cot[e_.+f_.*x_])^n_.,x_Symbol] :=
  g^(m+n)*Int[(g*Tan[e+f*x])^(p-m-n)*(b+a*Tan[e+f*x])^m*(d+c*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && Not[IntegerQ[p]] && IntegerQ[m] && IntegerQ[n]
```

2: $\int (g\,\mathrm{Tan}[e+f\,x]^q)^p\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $p\notin\mathbb{Z}\,\bigwedge\,\neg\,(m\in\mathbb{Z}\,\land\,n\in\mathbb{Z})$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(g\,\mathrm{Tan}[e+f\,x]^q)^p}{(g\,\mathrm{Tan}[e+f\,x])^{p\,q}} == 0$

Rule: If $p\notin\mathbb{Z}\,\bigwedge\,\neg\,(m\in\mathbb{Z}\,\land\,n\in\mathbb{Z})$**, then**

$$\int (g\,\mathrm{Tan}[e+f\,x]^q)^p\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx\;\longrightarrow\;\frac{(g\,\mathrm{Tan}[e+f\,x]^q)^p}{(g\,\mathrm{Tan}[e+f\,x])^{p\,q}}\int (g\,\mathrm{Tan}[e+f\,x])^{p\,q}\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx$$

Program code:

```
Int[(g_.*tan[e_.+f_.*x_]^q_)^p_*(a_.+b_.*tan[e_.+f_.*x_])^m_.*(c_+d_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  (g*Tan[e+f*x]^q)^p/(g*Tan[e+f*x])^(p*q)*Int[(g*Tan[e+f*x])^(p*q)*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p,q},x] && Not[IntegerQ[p]] && Not[IntegerQ[m] && IntegerQ[n]]
```

Rules for integrands of the form $(g\,\mathrm{Tan}[e+f\,x])^p\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Cot}[e+f\,x])^n$

1: $\int (g\,\mathrm{Tan}[e+f\,x])^p\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Cot}[e+f\,x])^n\,dx$ when $n\in\mathbb{Z}$

Derivation: Algebraic normalization

- **Basis:** $c+d\,\mathrm{Cot}[z] == \frac{d+c\,\mathrm{Tan}[z]}{\mathrm{Tan}[z]}$

Rule: If $n\in\mathbb{Z}$**, then**

$$\int (g\,\mathrm{Tan}[e+f\,x])^p\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Cot}[e+f\,x])^n\,dx\;\longrightarrow\;g^n\int (g\,\mathrm{Tan}[e+f\,x])^{p-n}\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(d+c\,\mathrm{Tan}[e+f\,x])^n\,dx$$

Program code:

```
Int[(g_.*tan[e_.+f_.*x_])^p_.*(a_+b_.*tan[e_.+f_.*x_])^m_.*(c_+d_.*cot[e_.+f_.*x_])^n_.,x_Symbol] :=
  g^n*Int[(g*Tan[e+f*x])^(p-n)*(a+b*Tan[e+f*x])^m*(d+c*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && IntegerQ[n]
```

2. $\int (g\,\text{Tan}[e+f\,x])^p\,(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Cot}[e+f\,x])^n\,dx$ when $n\notin\mathbb{Z}$

   1. $\int (g\,\text{Tan}[e+f\,x])^p\,(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Cot}[e+f\,x])^n\,dx$ when $n\notin\mathbb{Z}\,\land\,m\in\mathbb{Z}$

      1: $\int \text{Tan}[e+f\,x]^p\,(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Cot}[e+f\,x])^n\,dx$ when $n\notin\mathbb{Z}\,\bigwedge\,m\in\mathbb{Z}\,\bigwedge\,p\in\mathbb{Z}$

Derivation: Algebraic normalization

Basis: $a+b\,\text{Tan}[z]=\frac{b+a\,\text{Cot}[z]}{\text{Cot}[z]}$

Rule: If $n\notin\mathbb{Z}\,\bigwedge\,m\in\mathbb{Z}\,\bigwedge\,p\in\mathbb{Z}$**, then**

$$\int \text{Tan}[e+f\,x]^p\,(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Cot}[e+f\,x])^n\,dx\,\longrightarrow\,\int \frac{(b+a\,\text{Cot}[e+f\,x])^m\,(c+d\,\text{Cot}[e+f\,x])^n}{\text{Cot}[e+f\,x]^{m+p}}\,dx$$

Program code:

```
Int[tan[e_.+f_.*x_]^p_.*(a_+b_.*tan[e_.+f_.*x_])^m_.*(c_+d_.*cot[e_.+f_.*x_])^n_,x_Symbol] :=
  Int[(b+a*Cot[e+f*x])^m*(c+d*Cot[e+f*x])^n/Cot[e+f*x]^(m+p),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && Not[IntegerQ[n]] && IntegerQ[m] && IntegerQ[p]
```

      2: $\int (g\,\text{Tan}[e+f\,x])^p\,(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Cot}[e+f\,x])^n\,dx$ when $n\notin\mathbb{Z}\,\bigwedge\,m\in\mathbb{Z}\,\bigwedge\,p\notin\mathbb{Z}$

Derivation: Algebraic normalization and piecewise constant extraction

Basis: $a+b\,\text{Tan}[z]=\frac{b+a\,\text{Cot}[z]}{\text{Cot}[z]}$

Basis: $\partial_x\left(\text{Cot}[e+f\,x]^p\,(g\,\text{Tan}[e+f\,x])^p\right)=0$

Rule: If $n\notin\mathbb{Z}\,\bigwedge\,m\in\mathbb{Z}\,\bigwedge\,p\notin\mathbb{Z}$**, then**

$$\int (g\,\text{Tan}[e+f\,x])^p\,(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Cot}[e+f\,x])^n\,dx\,\longrightarrow\,\text{Cot}[e+f\,x]^p\,(g\,\text{Tan}[e+f\,x])^p\int \frac{(b+a\,\text{Cot}[e+f\,x])^m\,(c+d\,\text{Cot}[e+f\,x])^n}{\text{Cot}[e+f\,x]^{m+p}}\,dx$$

Program code:

```
Int[(g_.*tan[e_.+f_.*x_])^p_*(a_+b_.*tan[e_.+f_.*x_])^m_.*(c_+d_.*cot[e_.+f_.*x_])^n_,x_Symbol] :=
  Cot[e+f*x]^p*(g*Tan[e+f*x])^p*Int[(b+a*Cot[e+f*x])^m*(c+d*Cot[e+f*x])^n/Cot[e+f*x]^(m+p),x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && Not[IntegerQ[n]] && IntegerQ[m] && Not[IntegerQ[p]]
```

2: $\int (g\,\mathrm{Tan}[e+f\,x])^p\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Cot}[e+f\,x])^n\,dx$ when $n\notin\mathbb{Z}\land m\notin\mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \frac{(c+d\,\mathrm{Cot}[e+f\,x])^n\,(g\,\mathrm{Tan}[e+f\,x])^n}{(d+c\,\mathrm{Tan}[e+f\,x])^n} == 0$
- **Rule: If** $n\notin\mathbb{Z}\land m\notin\mathbb{Z}$**, then**

$$\int (g\,\mathrm{Tan}[e+f\,x])^p\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Cot}[e+f\,x])^n\,dx\longrightarrow$$
$$\frac{(g\,\mathrm{Tan}[e+f\,x])^n\,(c+d\,\mathrm{Cot}[e+f\,x])^n}{(d+c\,\mathrm{Tan}[e+f\,x])^n}\int (g\,\mathrm{Tan}[e+f\,x])^{p-n}\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(d+c\,\mathrm{Tan}[e+f\,x])^n\,dx$$

- **Program code:**

```
Int[(g_.*tan[e_.+f_.*x_])^p_.*(a_+b_.*tan[e_.+f_.*x_])^m_*(c_+d_.*cot[e_.+f_.*x_])^n_,x_Symbol] :=
  (g*Tan[e+f*x])^n*(c+d*Cot[e+f*x])^n/(d+c*Tan[e+f*x])^n*Int[(g*Tan[e+f*x])^(p-n)*(a+b*Tan[e+f*x])^m*(d+c*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && Not[IntegerQ[n]] && Not[IntegerQ[m]]
```