Rules for integrands of the form $u\,(a+b\,\text{ArcSinh}[c\,x])^n$

1. $\int (d+e\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$

   1. $\int (d+e\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

      1: $\int \frac{(a+b\,\text{ArcSinh}[c\,x])^n}{d+e\,x}\,dx$

- Derivation: Integration by substitution
- Basis: $\frac{1}{d+e\,x} = \text{Subst}\left[\frac{\text{Cosh}[x]}{c\,d+e\,\text{Sinh}[x]},\,x,\,\text{ArcSinh}[c\,x]\right]\,\partial_x\,\text{ArcSinh}[c\,x]$
- Note: $\frac{(a+b\,x)^n\,\text{Cosh}[x]}{c\,d+e\,\text{Sinh}[x]}$ is not integrable unless $n\in\mathbb{Z}^+$.
- Rule: If $n\in\mathbb{Z}^+$, then

$$\int \frac{(a+b\,\text{ArcSinh}[c\,x])^n}{d+e\,x}\,dx \longrightarrow \text{Subst}\left[\int \frac{(a+b\,x)^n\,\text{Cosh}[x]}{c\,d+e\,\text{Sinh}[x]}\,dx,\,x,\,\text{ArcSinh}[c\,x]\right]$$

- Program code:

```
Int[(a_.+b_.*ArcSinh[c_.*x_])^n_./(d_.+e_.*x_),x_Symbol] :=
  Subst[Int[(a+b*x)^n*Cosh[x]/(c*d+e*Sinh[x]),x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[n,0]
```

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./(d_.+e_.*x_),x_Symbol] :=
  Subst[Int[(a+b*x)^n*Sinh[x]/(c*d+e*Cosh[x]),x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[n,0]
```

2: $\int (d+e\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+\land m\neq -1$

- **Reference: G&R 2.831, CRC 453, A&S 4.4.65**
- **Reference: G&R 2.832, CRC 454, A&S 4.4.67**
- **Derivation: Integration by parts**
- **Basis: If** $m\neq -1$**, then** $(d+e\,x)^m = \partial_x \frac{(d+e\,x)^{m+1}}{e\,(m+1)}$
- **Rule: If** $n\in\mathbb{Z}^+\land m\neq -1$**, then**

$$\int (d+e\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow \frac{(d+e\,x)^{m+1}\,(a+b\,\text{ArcSinh}[c\,x])^n}{e\,(m+1)} - \frac{b\,c\,n}{e\,(m+1)}\int \frac{(d+e\,x)^{m+1}\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}}{\sqrt{1+c^2\,x^2}}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*ArcSinh[c*x])^n/(e*(m+1)) -
  b*c*n/(e*(m+1))*Int[(d+e*x)^(m+1)*(a+b*ArcSinh[c*x])^(n-1)/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[n,0] && NeQ[m,-1]
```

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*ArcCosh[c*x])^n/(e*(m+1)) -
  b*c*n/(e*(m+1))*Int[(d+e*x)^(m+1)*(a+b*ArcCosh[c*x])^(n-1)/(Sqrt[-1+c*x]*Sqrt[1+c*x]),x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[n,0] && NeQ[m,-1]
```

2. $\int (d+e x)^m (a+b \text{ArcSinh}[c x])^n \, dx$ when $m \in \mathbb{Z}^+$

1: $\int (d+e x)^m (a+b \text{ArcSinh}[c x])^n \, dx$ when $m \in \mathbb{Z}^+ \wedge n < -1$

Derivation: Algebraic expansion

Rule: If $m \in \mathbb{Z}^+ \wedge n < -1$**, then**

$$\int (d+e x)^m (a+b \text{ArcSinh}[c x])^n \, dx \longrightarrow \int \text{ExpandIntegrand}[(d+e x)^m (a+b \text{ArcSinh}[c x])^n, x] \, dx$$

Program code:

```
Int[(d_+e_.*x_)^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(a+b*ArcSinh[c*x])^n,x],x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[m,0] && LtQ[n,-1]
```

```
Int[(d_+e_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(a+b*ArcCosh[c*x])^n,x],x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[m,0] && LtQ[n,-1]
```

2: $\int (d+e x)^m (a+b \text{ArcSinh}[c x])^n \, dx$ when $m \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F[x] == \frac{1}{c} F\left[\frac{\text{Sinh}[\text{ArcSinh}[c x]]}{c}\right] \text{Cosh}[\text{ArcSinh}[c x]] \, \partial_x \text{ArcSinh}[c x]$

Note: If $m \in \mathbb{Z}^+$**, then** $(a+b x)^n \text{Cosh}[x] (c d+e \text{Sinh}[x])^m$ **is integrable in closed-form.**

Rule: If $m \in \mathbb{Z}^+$**, then**

$$\int (d+e x)^m (a+b \text{ArcSinh}[c x])^n \, dx \longrightarrow \frac{1}{c^{m+1}} \text{Subst}\left[\int (a+b x)^n \text{Cosh}[x] (c d+e \text{Sinh}[x])^m \, dx, x, \text{ArcSinh}[c x]\right]$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  1/c^(m+1)*Subst[Int[(a+b*x)^n*Cosh[x]*(c*d+e*Sinh[x])^m,x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[m,0]
```

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  1/c^(m+1)*Subst[Int[(a+b*x)^n*(c*d+e*Cosh[x])^m*Sinh[x],x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[m,0]
```

2. $\int P_x\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$

1: $\int P_x\,(a+b\,\text{ArcSinh}[c\,x])\,dx$

Derivation: Integration by parts

Rule: Let $u=\int P_x\,dx$**, then**

$$\int P_x\,(a+b\,\text{ArcSinh}[c\,x])\,dx\;\longrightarrow\;u\,(a+b\,\text{ArcSinh}[c\,x])-b\,c\int\frac{u}{\sqrt{1+c^2\,x^2}}\,dx$$

$$\int P_x\,(a+b\,\text{ArcCosh}[c\,x])\,dx\;\longrightarrow\;u\,(a+b\,\text{ArcCosh}[c\,x])-\frac{b\,c\,\sqrt{1-c^2\,x^2}}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}\int\frac{u}{\sqrt{1-c^2\,x^2}}\,dx$$

Program code:

```
Int[Px_*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[ExpandExpression[Px,x],x]},
  Dist[a+b*ArcSinh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/Sqrt[1+c^2*x^2],x],x]] /;
FreeQ[{a,b,c},x] && PolynomialQ[Px,x]
```

```
Int[Px_*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[ExpandExpression[Px,x],x]},
  Dist[a+b*ArcCosh[c*x],u,x] - b*c*Sqrt[1-c^2*x^2]/(Sqrt[-1+c*x]*Sqrt[1+c*x])*Int[SimplifyIntegrand[u/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c},x] && PolynomialQ[Px,x]
```

x: $\int P_x\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $n\in\mathbb{Z}^+$**, let** $u=\int P_x\,dx$**, then**

$$\int P_x\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx\;\longrightarrow\;u\,(a+b\,\text{ArcSinh}[c\,x])^n-b\,c\,n\int\frac{u\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}}{\sqrt{1+c^2\,x^2}}\,dx$$

$$\int P_x\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx \longrightarrow u\,(a+b\,\text{ArcCosh}[c\,x])^n-\frac{b\,c\,n\,\sqrt{1-c^2\,x^2}}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}\int\frac{u\,(a+b\,\text{ArcCosh}[c\,x])^{n-1}}{\sqrt{1-c^2\,x^2}}\,dx$$

Program code:

```
(* Int[Px_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  With[{u=IntHide[Px,x]},
  Dist[(a+b*ArcSinh[c*x])^n,u,x] - b*c*n*Int[SimplifyIntegrand[u*(a+b*ArcSinh[c*x])^(n-1)/Sqrt[1+c^2*x^2],x],x]] /;
FreeQ[{a,b,c},x] && PolynomialQ[Px,x] && IGtQ[n,0] *)
```

```
(* Int[Px_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  With[{u=IntHide[Px,x]},
  Dist[(a+b*ArcCosh[c*x])^n,u,x] -
    b*c*n*Sqrt[1-c^2*x^2]/(Sqrt[-1+c*x]*Sqrt[1+c*x])*Int[SimplifyIntegrand[u*(a+b*ArcCosh[c*x])^(n-1)/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c},x] && PolynomialQ[Px,x] && IGtQ[n,0] *)
```

2: $\int P_x\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $n\neq 1$

Derivation: Algebraic expansion

Rule: If $n\neq 1$, then

$$\int P_x\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow \int \text{ExpandIntegrand}[P_x\,(a+b\,\text{ArcSinh}[c\,x])^n,\,x]\,dx$$

Program code:

```
Int[Px_*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[Px*(a+b*ArcSinh[c*x])^n,x],x] /;
FreeQ[{a,b,c,n},x] && PolynomialQ[Px,x]
```

```
Int[Px_*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[Px*(a+b*ArcCosh[c*x])^n,x],x] /;
FreeQ[{a,b,c,n},x] && PolynomialQ[Px,x]
```

3. $\int P_x (d+e x)^m (a+b \text{ArcSinh}[c x])^n \, dx$ when $n \in \mathbb{Z}^+$

1: $\int P_x (d+e x)^m (a+b \text{ArcSinh}[c x]) \, dx$

Derivation: Integration by parts

Rule: Let $u = \int P_x (d+e x)^m \, dx$**, then**

$$\int P_x (d+e x)^m (a+b \text{ArcSinh}[c x]) \, dx \longrightarrow u (a+b \text{ArcSinh}[c x]) - b c \int \frac{u}{\sqrt{1+c^2 x^2}} \, dx$$

$$\int P_x (d+e x)^m (a+b \text{ArcCosh}[c x]) \, dx \longrightarrow u (a+b \text{ArcCosh}[c x]) - \frac{b c \sqrt{1-c^2 x^2}}{\sqrt{-1+c x} \sqrt{1+c x}} \int \frac{u}{\sqrt{1-c^2 x^2}} \, dx$$

Program code:

```
Int[Px_*(d_.+e_.*x_)^m_.*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[Px*(d+e*x)^m,x]},
  Dist[a+b*ArcSinh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/Sqrt[1+c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e,m},x] && PolynomialQ[Px,x]
```

```
Int[Px_*(d_.+e_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[Px*(d+e*x)^m,x]},
  Dist[a+b*ArcCosh[c*x],u,x] - b*c*Sqrt[1-c^2*x^2]/(Sqrt[-1+c*x]*Sqrt[1+c*x])*Int[SimplifyIntegrand[u/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e,m},x] && PolynomialQ[Px,x]
```

2: $\int (f+g\,x)^p\,(d+e\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $(n\mid p)\in\mathbb{Z}^+\land m\in\mathbb{Z}^-\land m+p+1<0$

Derivation: Integration by parts

Note: If $p\in\mathbb{Z}^+\land m\in\mathbb{Z}^-\land m+p+1<0$, then $\int (f+g\,x)^p\,(d+e\,x)^m\,dx$ is a rational function.

Rule: If $(n\mid p)\in\mathbb{Z}^+\land m\in\mathbb{Z}^-\land m+p+1<0$, let $u=\int (f+g\,x)^p\,(d+e\,x)^m\,dx$, then

$$\int (f+g\,x)^p\,(d+e\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx\;\longrightarrow\;u\,(a+b\,\text{ArcSinh}[c\,x])^n-b\,c\,n\int\frac{u\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}}{\sqrt{1+c^2\,x^2}}\,dx$$

Program code:

```
Int[(f_.+g_.*x_)^p_.*(d_+e_.*x_)^m_*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  With[{u=IntHide[(f+g*x)^p*(d+e*x)^m,x]},
  Dist[(a+b*ArcSinh[c*x])^n,u,x] - b*c*n*Int[SimplifyIntegrand[u*(a+b*ArcSinh[c*x])^(n-1)/Sqrt[1+c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[n,0] && IGtQ[p,0] && ILtQ[m,0] && LtQ[m+p+1,0]
```

```
Int[(f_.+g_.*x_)^p_.*(d_+e_.*x_)^m_*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  With[{u=IntHide[(f+g*x)^p*(d+e*x)^m,x]},
  Dist[(a+b*ArcCosh[c*x])^n,u,x] - b*c*n*Int[SimplifyIntegrand[u*(a+b*ArcCosh[c*x])^(n-1)/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[n,0] && IGtQ[p,0] && ILtQ[m,0] && LtQ[m+p+1,0]
```

3: $\int \frac{(f+g x+h x^2)^p (a+b \text{ArcSinh}[c x])^n}{(d+e x)^2} dx$ when $(n \mid p) \in \mathbb{Z}^+ \wedge e g-2 d h == 0$

Derivation: Integration by parts

- **Note: If** $p \in \mathbb{Z}^+ \wedge e g-2 d h == 0$**, then** $\int \frac{(f+g x+h x^2)^p}{(d+e x)^2} dx$ **is a rational function.**

- **Rule: If** $(n \mid p) \in \mathbb{Z}^+ \wedge e g-2 d h == 0$**, let** $u = \int \frac{(f+g x+h x^2)^p}{(d+e x)^2} dx$**, then**

$$\int \frac{(f+g x+h x^2)^p (a+b \text{ArcSinh}[c x])^n}{(d+e x)^2} dx \longrightarrow u (a+b \text{ArcSinh}[c x])^n - b c n \int \frac{u (a+b \text{ArcSinh}[c x])^{n-1}}{\sqrt{1+c^2 x^2}} dx$$

Program code:

```
Int[(f_.+g_.*x_+h_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_/(d_+e_.*x_)^2,x_Symbol] :=
  With[{u=IntHide[(f+g*x+h*x^2)^p/(d+e*x)^2,x]},
  Dist[(a+b*ArcSinh[c*x])^n,u,x] - b*c*n*Int[SimplifyIntegrand[u*(a+b*ArcSinh[c*x])^(n-1)/Sqrt[1+c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && IGtQ[n,0] && IGtQ[p,0] && EqQ[e*g-2*d*h,0]
```

```
Int[(f_.+g_.*x_+h_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_/(d_+e_.*x_)^2,x_Symbol] :=
  With[{u=IntHide[(f+g*x+h*x^2)^p/(d+e*x)^2,x]},
  Dist[(a+b*ArcCosh[c*x])^n,u,x] - b*c*n*Int[SimplifyIntegrand[u*(a+b*ArcCosh[c*x])^(n-1)/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x],x]] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && IGtQ[n,0] && IGtQ[p,0] && EqQ[e*g-2*d*h,0]
```

4: $\int P_x (d+e x)^m (a+b \text{ArcSinh}[c x])^n dx$ when $n \in \mathbb{Z}^+ \wedge m \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $n \in \mathbb{Z}^+ \wedge m \in \mathbb{Z}$**, then**

$$\int P_x (d+e x)^m (a+b \text{ArcSinh}[c x])^n dx \longrightarrow \int \text{ExpandIntegrand}[P_x (d+e x)^m (a+b \text{ArcSinh}[c x])^n, x] dx$$

Program code:

```
Int[Px_*(d_+e_.*x_)^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[Px*(d+e*x)^m*(a+b*ArcSinh[c*x])^n,x],x] /;
FreeQ[{a,b,c,d,e},x] && PolynomialQ[Px,x] && IGtQ[n,0] && IntegerQ[m]
```

```
Int[Px_*(d_+e_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[Px*(d+e*x)^m*(a+b*ArcCosh[c*x])^n,x],x] /;
FreeQ[{a,b,c,d,e},x] && PolynomialQ[Px,x] && IGtQ[n,0] && IntegerQ[m]
```

4. $\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e = c^2 d\ \bigwedge\ m \in \mathbb{Z}\ \bigwedge\ p-\frac{1}{2} \in \mathbb{Z}$

   1. $\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e = c^2 d\ \bigwedge\ m \in \mathbb{Z}\ \bigwedge\ p-\frac{1}{2} \in \mathbb{Z}\ \bigwedge\ d>0$

      1: $\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])\,dx$ when $e = c^2 d\ \bigwedge\ m \in \mathbb{Z}\ \bigwedge\ p+\frac{1}{2} \in \mathbb{Z}^-\ \bigwedge\ d>0\ \bigwedge\ m>0$

Derivation: Integration by parts

Note: If $m \in \mathbb{Z}\ \bigwedge\ p+\frac{1}{2} \in \mathbb{Z}\ \bigwedge\ 0<m<-2p-1$**, then** $\int (f+g\,x)^m \left(d+e\,x^2\right)^p dx$ **is an algebraic function.**

Rule: If $e = c^2 d\ \bigwedge\ m \in \mathbb{Z}\ \bigwedge\ p+\frac{1}{2} \in \mathbb{Z}^-\ \bigwedge\ d>0\ \bigwedge\ m>0$**, let** $u = \int (f+g\,x)^m \left(d+e\,x^2\right)^p dx$**, then**

$$\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])\,dx \longrightarrow u\,(a+b\,\text{ArcSinh}[c\,x]) - b\,c \int \frac{u}{\sqrt{1+c^2 x^2}}\,dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f+g*x)^m*(d+e*x^2)^p,x]},
  Dist[a+b*ArcSinh[c*x],u,x] - b*c*Int[Dist[1/Sqrt[1+c^2*x^2],u,x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[e,c^2*d] && IGtQ[m,0] && ILtQ[p+1/2,0] && GtQ[d,0] && (LtQ[m,-2*p-1] || GtQ[m,3])
```

```
Int[(f_+g_.*x_)^m_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f+g*x)^m*(d1+e1*x)^p*(d2+e2*x)^p,x]},
  Dist[a+b*ArcCosh[c*x],u,x] - b*c*Int[Dist[1/(Sqrt[1+c*x]*Sqrt[-1+c*x]),u,x],x]] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IGtQ[m,0] && ILtQ[p+1/2,0] && GtQ[d1,0] && LtQ[d2,0] &&
  (LtQ[m,-2*p-1] || GtQ[m,3])
```

2: $\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2 d \bigwedge m \in \mathbb{Z} \bigwedge p+\frac{1}{2} \in \mathbb{Z} \bigwedge d>0 \bigwedge n \in \mathbb{Z}^+ \bigwedge m>0$

Derivation: Algebraic expansion

Rule: If $e == c^2 d \bigwedge m \in \mathbb{Z} \bigwedge p+\frac{1}{2} \in \mathbb{Z} \bigwedge d>0 \bigwedge n \in \mathbb{Z}^+ \bigwedge m>0$**, then**

$$\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow \int \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,\text{ExpandIntegrand}\left[(f+g\,x)^m,\,x\right]dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n,(f+g*x)^m,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[e,c^2*d] && IGtQ[m,0] && IntegerQ[p+1/2] && GtQ[d,0] && IGtQ[n,0] &&
  (EqQ[n,1] && GtQ[p,-1] || GtQ[p,0] || EqQ[m,1] || EqQ[m,2] && LtQ[p,-2])
```

```
Int[(f_+g_.*x_)^m_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n,(f+g*x)^m,x],x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IGtQ[m,0] && IntegerQ[p+1/2] && GtQ[d1,0] && LtQ[d2,0] && I
  (EqQ[n,1] && GtQ[p,-1] || GtQ[p,0] || EqQ[m,1] || EqQ[m,2] && LtQ[p,-2])
```

3. $\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2 d \bigwedge m \in \mathbb{Z} \bigwedge p+\frac{1}{2} \in \mathbb{Z}^+ \bigwedge d>0$

1: $\int (f+g\,x)^m \sqrt{d+e\,x^2}\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2 d \wedge m \in \mathbb{Z} \wedge d>0 \wedge n \in \mathbb{Z}^+ \wedge m<0$

Derivation: Integration by parts

Basis: If $e == c^2 d \wedge d>0$**, then** $\frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}} == \partial_x \frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,\sqrt{d}\,(n+1)}$

Rule: If $e == c^2 d \wedge m \in \mathbb{Z} \wedge d>0 \wedge n \in \mathbb{Z}^+ \wedge m<0$**, then**

$$\int (f+g\,x)^m \sqrt{d+e\,x^2}\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow$$

$$\frac{(f+g\,x)^m \left(d+e\,x^2\right) (a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,\sqrt{d}\,(n+1)} -$$

$$\frac{1}{b\,c\,\sqrt{d}\,(n+1)} \int \left(d\,g\,m+2\,e\,f\,x+e\,g\,(m+2)\,x^2\right) (f+g\,x)^{m-1} (a+b\,\text{ArcSinh}[c\,x])^{n+1}\,dx$$

Program code:

```
Int[(f_.+g_.*x_)^m_*Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f+g*x)^m*(d+e*x^2)*(a+b*ArcSinh[c*x])^(n+1)/(b*c*Sqrt[d]*(n+1)) -
  1/(b*c*Sqrt[d]*(n+1))*Int[(d*g*m+2*e*f*x+e*g*(m+2)*x^2)*(f+g*x)^(m-1)*(a+b*ArcSinh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[e,c^2*d] && ILtQ[m,0] && GtQ[d,0] && IGtQ[n,0]
```

```
Int[(f_+g_.*x_)^m_*Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f+g*x)^m*(d1*d2+e1*e2*x^2)*(a+b*ArcCosh[c*x])^(n+1)/(b*c*Sqrt[-d1*d2]*(n+1)) -
  1/(b*c*Sqrt[-d1*d2]*(n+1))*Int[(d1*d2*g*m+2*e1*e2*f*x+e1*e2*g*(m+2)*x^2)*(f+g*x)^(m-1)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && ILtQ[m,0] && GtQ[d1,0] && LtQ[d2,0] && IGtQ[n,0]
```

2: $\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e=c^2\,d\ \bigwedge\ m\in\mathbb{Z}\ \bigwedge\ p+\frac{1}{2}\in\mathbb{Z}^+\ \bigwedge\ d>0\ \bigwedge\ n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

- **Rule: If** $e=c^2\,d\ \bigwedge\ m\in\mathbb{Z}\ \bigwedge\ p+\frac{1}{2}\in\mathbb{Z}^+\ \bigwedge\ d>0\ \bigwedge\ n\in\mathbb{Z}^+$**, then**

$$\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow \int \sqrt{d+e\,x^2}\,(a+b\,\text{ArcSinh}[c\,x])^n\,\text{ExpandIntegrand}\left[(f+g\,x)^m \left(d+e\,x^2\right)^{p-1/2},\,x\right]dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[Sqrt[d+e*x^2]*(a+b*ArcSinh[c*x])^n,(f+g*x)^m*(d+e*x^2)^(p-1/2),x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[e,c^2*d] && IntegerQ[m] && IGtQ[p+1/2,0] && GtQ[d,0] && IGtQ[n,0]
```

```
Int[(f_+g_.*x_)^m_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[Sqrt[d1+e1*x]*Sqrt[d2+e2*x]*(a+b*ArcCosh[c*x])^n,(f+g*x)^m*(d1+e1*x)^(p-1/2)*(d2+e2*x)^(p-1/2),x],x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[m] && IGtQ[p+1/2,0] && GtQ[d1,0] && LtQ[d2,0] && I
```

3: $\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2 d \bigwedge m \in \mathbb{Z} \bigwedge p-\frac{1}{2} \in \mathbb{Z}^+ \bigwedge d>0 \bigwedge n \in \mathbb{Z}^+ \bigwedge m<0$

Derivation: Integration by parts

- **Basis: If** $e == c^2 d \wedge d>0$, **then** $\frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}} == \partial_x \frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,\sqrt{d}\,(n+1)}$
- **Rule: If** $e == c^2 d \bigwedge m \in \mathbb{Z} \bigwedge p-\frac{1}{2} \in \mathbb{Z}^+ \bigwedge d>0 \bigwedge n \in \mathbb{Z}^+ \bigwedge m<0$, **then**

$$\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow$$

$$\frac{(f+g\,x)^m \left(d+e\,x^2\right)^{p+\frac{1}{2}} (a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,\sqrt{d}\,(n+1)} -$$

$$\frac{1}{b\,c\,\sqrt{d}\,(n+1)} \int (f+g\,x)^{m-1} (a+b\,\text{ArcSinh}[c\,x])^{n+1}\,\text{ExpandIntegrand}\left[\left(d\,g\,m+e\,f\,(2\,p+1)\,x+e\,g\,(m+2\,p+1)\,x^2\right) \left(d+e\,x^2\right)^{p-\frac{1}{2}},\,x\right] dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f+g*x)^m*(d+e*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n+1)/(b*c*Sqrt[d]*(n+1)) -
  1/(b*c*Sqrt[d]*(n+1))*
    Int[ExpandIntegrand[(f+g*x)^(m-1)*(a+b*ArcSinh[c*x])^(n+1),(d*g*m+e*f*(2*p+1)*x+e*g*(m+2*p+1)*x^2)*(d+e*x^2)^(p-1/2),x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[e,c^2*d] && ILtQ[m,0] && IGtQ[p-1/2,0] && GtQ[d,0] && IGtQ[n,0]
```

```
Int[(f_+g_.*x_)^m_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f+g*x)^m*(d1+e1*x)^(p+1/2)*(d2+e2*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n+1)/(b*c*Sqrt[-d1*d2]*(n+1)) -
  1/(b*c*Sqrt[-d1*d2]*(n+1))*
    Int[ExpandIntegrand[(f+g*x)^(m-1)*(a+b*ArcCosh[c*x])^(n+1),
      (d1*d2*g*m+e1*e2*f*(2*p+1)*x+e1*e2*g*(m+2*p+1)*x^2)*(d1+e1*x)^(p-1/2)*(d2+e2*x)^(p-1/2),x],x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && ILtQ[m,0] && IGtQ[p-1/2,0] && GtQ[d1,0] && LtQ[d2,0] && IGt
```

4. $\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d \bigwedge m \in \mathbb{Z} \bigwedge p-\frac{1}{2} \in \mathbb{Z}^- \bigwedge d>0$

1. $\int \frac{(f+g\,x)^m (a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2\,d \wedge m \in \mathbb{Z} \wedge d>0$

1: $\int \frac{(f+g\,x)^m (a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2\,d \wedge m \in \mathbb{Z} \wedge d>0 \wedge m>0 \wedge n<-1$

- **Derivation: Integration by parts**

- **Basis: If** $e == c^2\,d \wedge d>0$**, then** $\frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}} == \partial_x \frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,\sqrt{d}\,(n+1)}$

- **Rule: If** $e == c^2\,d \wedge m \in \mathbb{Z} \wedge d>0 \wedge m>0 \wedge n<-1$**, then**

$$\int \frac{(f+g\,x)^m (a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx \longrightarrow \frac{(f+g\,x)^m (a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,\sqrt{d}\,(n+1)} - \frac{g\,m}{b\,c\,\sqrt{d}\,(n+1)} \int (f+g\,x)^{m-1} (a+b\,\text{ArcSinh}[c\,x])^{n+1}\,dx$$

- **Program code:**

```
Int[(f_+g_.*x_)^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  (f+g*x)^m*(a+b*ArcSinh[c*x])^(n+1)/(b*c*Sqrt[d]*(n+1)) -
  g*m/(b*c*Sqrt[d]*(n+1))*Int[(f+g*x)^(m-1)*(a+b*ArcSinh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[e,c^2*d] && IGtQ[m,0] && GtQ[d,0] && LtQ[n,-1]
```

```
Int[(f_+g_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_/(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  (f+g*x)^m*(a+b*ArcCosh[c*x])^(n+1)/(b*c*Sqrt[-d1*d2]*(n+1)) -
  g*m/(b*c*Sqrt[-d1*d2]*(n+1))*Int[(f+g*x)^(m-1)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IGtQ[m,0] && GtQ[d1,0] && LtQ[d2,0] && LtQ[n,-1]
```

2: $\int \frac{(f+g x)^m (a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{d+e x^2}} \, dx$ when $e == c^2 d \wedge m \in \mathbb{Z} \wedge d > 0 \wedge (m > 0 \vee n \in \mathbb{Z}^+)$

Derivation: Integration by substitution

- **Basis: If** $e == c^2 d \wedge d > 0$, **then** $\frac{F[x]}{\sqrt{d+e x^2}} == \frac{1}{c \sqrt{d}} \operatorname{Subst}\left[F\left[\frac{\operatorname{Sinh}[x]}{c}\right], x, \operatorname{ArcSinh}[c x]\right] \partial_x \operatorname{ArcSinh}[c x]$
- **Basis: If** $d_1 > 0 \wedge d_2 < 0$, **then** $\frac{F[x]}{\sqrt{d_1+c d_1 x} \sqrt{d_2-c d_2 x}} == \frac{1}{c \sqrt{-d_1 d_2}} \operatorname{Subst}\left[F\left[\frac{\operatorname{Cosh}[x]}{c}\right], x, \operatorname{ArcCosh}[c x]\right] \partial_x \operatorname{ArcCosh}[c x]$

Note: *Mathematica* 8 is unable to validate antiderivatives of ArcCosh rule when c is symbolic.

Rule: If $e == c^2 d \wedge m \in \mathbb{Z} \wedge d > 0 \wedge (m > 0 \vee n \in \mathbb{Z}^+)$, **then**

$$\int \frac{(f+g x)^m (a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{d+e x^2}} \, dx \to \frac{1}{c^{m+1} \sqrt{d}} \operatorname{Subst}\left[\int (a+b x)^n (c f+g \operatorname{Sinh}[x])^m \, dx, x, \operatorname{ArcSinh}[c x]\right]$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  1/(c^(m+1)*Sqrt[d])*Subst[Int[(a+b*x)^n*(c*f+g*Sinh[x])^m,x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[e,c^2*d] && IntegerQ[m] && GtQ[d,0] && (GtQ[m,0] || IGtQ[n,0])
```

```
Int[(f_+g_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_./(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  1/(c^(m+1)*Sqrt[-d1*d2])*Subst[Int[(a+b*x)^n*(c*f+g*Cosh[x])^m,x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g,n},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[m] && GtQ[d1,0] && LtQ[d2,0] && (GtQ[m,0] || IGt
```

2: $\int (f+g x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \wedge m \in \mathbb{Z} \wedge p+\frac{1}{2} \in \mathbb{Z}^- \wedge d > 0 \wedge n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

- **Rule: If** $e == c^2 d \wedge m \in \mathbb{Z} \wedge p+\frac{1}{2} \in \mathbb{Z}^- \wedge d > 0 \wedge n \in \mathbb{Z}^+$, **then**

$$\int (f+g x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx \to \int \frac{(a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{d+e x^2}} \operatorname{ExpandIntegrand}\left[(f+g x)^m \left(d+e x^2\right)^{p+1/2}, x\right] dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcSinh[c*x])^n/Sqrt[d+e*x^2],(f+g*x)^m*(d+e*x^2)^(p+1/2),x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[e,c^2*d] && IntegerQ[m] && ILtQ[p+1/2,0] && GtQ[d,0] && IGtQ[n,0]
```

```
Int[(f_+g_.*x_)^m_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCosh[c*x])^n/(Sqrt[d1+e1*x]*Sqrt[d2+e2*x]),(f+g*x)^m*(d1+e1*x)^(p+1/2)*(d2+e2*x)^(p+1/2),x],x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[m] && ILtQ[p+1/2,0] && GtQ[d1,0] && LtQ[d2,0] && I
```

2: $\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2 d \bigwedge m \in \mathbb{Z} \bigwedge p-\frac{1}{2} \in \mathbb{Z} \bigwedge d \not> 0$

Derivation: Piecewise constant extraction

- **Basis: If** $e == c^2 d$, **then** $\partial_x \frac{(d+e\,x^2)^p}{(1+c^2\,x^2)^p} == 0$
- **Rule: If** $e == c^2 d \bigwedge m \in \mathbb{Z} \bigwedge p-\frac{1}{2} \in \mathbb{Z} \bigwedge d \not> 0$, **then**

$$\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow \frac{d^{\text{IntPart}[p]} \left(d+e\,x^2\right)^{\text{FracPart}[p]}}{\left(1+c^2\,x^2\right)^{\text{FracPart}[p]}} \int (f+g\,x)^m \left(1+c^2\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  d^IntPart[p]*(d+e*x^2)^FracPart[p]/(1+c^2*x^2)^FracPart[p]*Int[(f+g*x)^m*(1+c^2*x^2)^p*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[e,c^2*d] && IntegerQ[m] && IntegerQ[p-1/2] && Not[GtQ[d,0]]
```

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d)^IntPart[p]*(d+e*x^2)^FracPart[p]/((1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f+g*x)^m*(1+c*x)^p*(-1+c*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[c^2*d+e,0] && IntegerQ[m] && IntegerQ[p-1/2]
```

```
Int[(f_+g_.*x_)^m_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(1-c^2*x^2)^FracPart[p]*
    Int[(f+g*x)^m*(1+c*x)^p*(-1+c*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g,n},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[m] && IntegerQ[p-1/2] && Not[GtQ[d1,0] && LtQ[d2
```

5. $\int \text{Log}[h\,(f+g\,x)^m]\,\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\bigwedge\,p-\frac{1}{2}\in\mathbb{Z}$

1. $\int \text{Log}[h\,(f+g\,x)^m]\,\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\bigwedge\,p-\frac{1}{2}\in\mathbb{Z}\,\bigwedge\,d>0$

1: $\int \frac{\text{Log}[h\,(f+g\,x)^m]\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2\,d\,\wedge\,d>0\,\wedge\,n\in\mathbb{Z}^+$

Derivation: Integration by parts

- **Basis: If** $e == c^2\,d\,\wedge\,d>0$**, then** $\frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}} == \partial_x\,\frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,\sqrt{d}\,(n+1)}$

- **Note: If** $n\in\mathbb{Z}^+$**, then** $\frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{f+g\,x}$ **is integrable in closed-form.**

- **Rule: If** $e == c^2\,d\,\wedge\,d>0\,\wedge\,n\in\mathbb{Z}^+$**, then**

$$\int \frac{\text{Log}[h\,(f+g\,x)^m]\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx \longrightarrow \frac{\text{Log}[h\,(f+g\,x)^m]\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,\sqrt{d}\,(n+1)} - \frac{g\,m}{b\,c\,\sqrt{d}\,(n+1)}\int \frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{f+g\,x}\,dx$$

Program code:

```
Int[Log[h_.*(f_.+g_.*x_)^m_.]*(a_.+b_.*ArcSinh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  Log[h*(f+g*x)^m]*(a+b*ArcSinh[c*x])^(n+1)/(b*c*Sqrt[d]*(n+1)) -
  g*m/(b*c*Sqrt[d]*(n+1))*Int[(a+b*ArcSinh[c*x])^(n+1)/(f+g*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,m},x] && EqQ[e,c^2*d] && GtQ[d,0] && IGtQ[n,0]
```

```
Int[Log[h_.*(f_.+g_.*x_)^m_.]*(a_.+b_.*ArcCosh[c_.*x_])^n_./(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  Log[h*(f+g*x)^m]*(a+b*ArcCosh[c*x])^(n+1)/(b*c*Sqrt[-d1*d2]*(n+1)) -
  g*m/(b*c*Sqrt[-d1*d2]*(n+1))*Int[(a+b*ArcCosh[c*x])^(n+1)/(f+g*x),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g,h,m},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[d1,0] && LtQ[d2,0] && IGtQ[n,0]
```

2: $\int \text{Log}[h\,(f+g\,x)^m]\,\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\bigwedge\,p-\frac{1}{2}\in\mathbb{Z}\,\bigwedge\,d\not>0$

Derivation: Piecewise constant extraction

- **Basis: If** $e == c^2\,d$**, then** $\partial_x\,\frac{(d+e\,x^2)^p}{(1+c^2\,x^2)^p} == 0$

- **Rule: If** $e == c^2\,d\,\bigwedge\,p-\frac{1}{2}\in\mathbb{Z}\,\bigwedge\,d\not>0$**, then**

$$\int \text{Log}[h\,(f+g\,x)^m]\,\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow \frac{d^{\text{IntPart}[p]}\,\left(d+e\,x^2\right)^{\text{FracPart}[p]}}{\left(1+c^2\,x^2\right)^{\text{FracPart}[p]}}\int \text{Log}[h\,(f+g\,x)^m]\,\left(1+c^2\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$$

Program code:

```
Int[Log[h_.*(f_.+g_.*x_)^m_.]*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  d^IntPart[p]*(d+e*x^2)^FracPart[p]/(1+c^2*x^2)^FracPart[p]*Int[Log[h*(f+g*x)^m]*(1+c^2*x^2)^p*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && EqQ[e,c^2*d] && IntegerQ[p-1/2] && Not[GtQ[d,0]]
```

```
Int[Log[h_.*(f_.+g_.*x_)^m_.]*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d)^IntPart[p]*(d+e*x^2)^FracPart[p]/((1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[Log[h*(f+g*x)^m]*(1+c*x)^p*(-1+c*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && EqQ[c^2*d+e,0] && IntegerQ[p-1/2]
```

```
Int[Log[h_.*(f_.+g_.*x_)^m_.]*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/((1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[Log[h*(f+g*x)^m]*(1+c*x)^p*(-1+c*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g,h,m,n},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[p-1/2] && Not[GtQ[d1,0] && LtQ[d2,0]]
```

6. $\int (d+e\,x)^m\,(f+g\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$

1: $\int (d+e\,x)^m\,(f+g\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])\,dx$ when $m+\frac{1}{2}\in\mathbb{Z}^-$

Derivation: Integration by parts

Rule: If $m+\frac{1}{2}\in\mathbb{Z}^-$**, let** $u=\int (d+e\,x)^m\,(f+g\,x)^m\,dx$**, then**

$$\int (d+e\,x)^m\,(f+g\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])\,dx \longrightarrow u\,(a+b\,\text{ArcSinh}[c\,x])-b\,c\int \frac{u}{\sqrt{1+c^2\,x^2}}\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_+g_.*x_)^m_*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x)^m*(f+g*x)^m,x]},
  Dist[a+b*ArcSinh[c*x],u,x] - b*c*Int[Dist[1/Sqrt[1+c^2*x^2],u,x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && ILtQ[m+1/2,0]
```

```
Int[(d_+e_.*x_)^m_*(f_+g_.*x_)^m_*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x)^m*(f+g*x)^m,x]},
  Dist[a+b*ArcCosh[c*x],u,x] - b*c*Int[Dist[1/(Sqrt[1+c*x]*Sqrt[-1+c*x]),u,x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && ILtQ[m+1/2,0]
```

2: $\int (d+e\,x)^m\,(f+g\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $m\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $m\in\mathbb{Z}$**, then**

$$\int (d+e\,x)^m\,(f+g\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow \int (a+b\,\text{ArcSinh}[c\,x])^n\,\text{ExpandIntegrand}[(d+e\,x)^m\,(f+g\,x)^m,\,x]\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_.*(f_+g_.*x_)^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcSinh[c*x])^n,(d+e*x)^m*(f+g*x)^m,x],x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && IntegerQ[m]
```

```
Int[(d_+e_.*x_)^m_.*(f_+g_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCosh[c*x])^n,(d+e*x)^m*(f+g*x)^m,x],x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && IntegerQ[m]
```

7: $\int u\,(a+b\,\text{ArcSinh}[c\,x])\,dx$ when $\int u\,dx$ is free of inverse functions

Derivation: Integration by parts

Rule: Let $v = \int u\,dx$, if v is free of inverse functions, then

$$\int u\,(a+b\,\text{ArcSinh}[c\,x])\,dx \longrightarrow v\,(a+b\,\text{ArcSinh}[c\,x]) - b\,c\int\frac{v}{\sqrt{1+c^2\,x^2}}\,dx$$

$$\int u\,(a+b\,\text{ArcCosh}[c\,x])\,dx \longrightarrow v\,(a+b\,\text{ArcCosh}[c\,x]) - \frac{b\,c\,\sqrt{1-c^2\,x^2}}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}\int\frac{v}{\sqrt{1-c^2\,x^2}}\,dx$$

Program code:

```
Int[u_*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{v=IntHide[u,x]},
  Dist[a+b*ArcSinh[c*x],v,x] - b*c*Int[SimplifyIntegrand[v/Sqrt[1+c^2*x^2],x],x] /;
 InverseFunctionFreeQ[v,x]] /;
FreeQ[{a,b,c},x]
```

```
Int[u_*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{v=IntHide[u,x]},
  Dist[a+b*ArcCosh[c*x],v,x] - b*c*Sqrt[1-c^2*x^2]/(Sqrt[-1+c*x]*Sqrt[1+c*x])*Int[SimplifyIntegrand[v/Sqrt[1-c^2*x^2],x],x] /;
 InverseFunctionFreeQ[v,x]] /;
FreeQ[{a,b,c},x]
```

8. $\int P_x\, u\, (a + b\, \text{ArcSinh}[c\, x])^n\, dx$

1: $\int P_x \left(d + e\, x^2\right)^p (a + b\, \text{ArcSinh}[c\, x])^n\, dx$ when $e == c^2 d \bigwedge p - \frac{1}{2} \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $e == c^2 d \bigwedge p - \frac{1}{2} \in \mathbb{Z}$, then

$$\int P_x \left(d + e\, x^2\right)^p (a + b\, \text{ArcSinh}[c\, x])^n\, dx \longrightarrow \int \text{ExpandIntegrand}\left[P_x \left(d + e\, x^2\right)^p (a + b\, \text{ArcSinh}[c\, x])^n,\, x\right] dx$$

Program code:

```
Int[Px_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  With[{u=ExpandIntegrand[Px*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,n},x] && PolynomialQ[Px,x] && EqQ[e,c^2*d] && IntegerQ[p-1/2]
```

```
Int[Px_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  With[{u=ExpandIntegrand[Px*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && PolynomialQ[Px,x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[p-1/2]
```

2: $\int P_x \left(f + g \left(d + e\, x^2\right)^p\right)^m (a + b\, \text{ArcSinh}[c\, x])^n\, dx$ when $e == c^2 d \bigwedge p + \frac{1}{2} \in \mathbb{Z}^+ \bigwedge (m \mid n) \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $e == c^2 d \bigwedge p + \frac{1}{2} \in \mathbb{Z}^+ \bigwedge (m \mid n) \in \mathbb{Z}$, then

$$\int P_x \left(f + g \left(d + e\, x^2\right)^p\right)^m (a + b\, \text{ArcSinh}[c\, x])^n\, dx \longrightarrow \int \text{ExpandIntegrand}\left[P_x \left(f + g \left(d + e\, x^2\right)^p\right)^m (a + b\, \text{ArcSinh}[c\, x])^n,\, x\right] dx$$

Program code:

```
Int[Px_.*(f_+g_.*(d_+e_.*x_^2)^p_)^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[Px*(f+g*(d+e*x^2)^p)^m*(a+b*ArcSinh[c*x])^n,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,f,g},x] && PolynomialQ[Px,x] && EqQ[e,c^2*d] && IGtQ[p+1/2,0] && IntegersQ[m,n]
```

```
Int[Px_.*(f_+g_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[Px*(f+g*(d1+e1*x)^p*(d2+e2*x)^p)^m*(a+b*ArcCosh[c*x])^n,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && PolynomialQ[Px,x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IGtQ[p+1/2,0] && IntegersQ[m,n]
```

9. $\int RF_x\, u\, (a + b\, \text{ArcSinh}[c\, x])^n \, dx$ when $n \in \mathbb{Z}^+$

1. $\int RF_x\, (a + b\, \text{ArcSinh}[c\, x])^n \, dx$ when $n \in \mathbb{Z}^+$

1: $\int RF_x\, \text{ArcSinh}[c\, x]^n \, dx$ when $n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n \in \mathbb{Z}^+$, then

$$\int RF_x\, \text{ArcSinh}[c\, x]^n \, dx \longrightarrow \int \text{ArcSinh}[c\, x]^n\, \text{ExpandIntegrand}[RF_x,\, x] \, dx$$

Program code:

```
Int[RFx_*ArcSinh[c_.*x_]^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[ArcSinh[c*x]^n,RFx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[c,x] && RationalFunctionQ[RFx,x] && IGtQ[n,0]
```

```
Int[RFx_*ArcCosh[c_.*x_]^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[ArcCosh[c*x]^n,RFx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[c,x] && RationalFunctionQ[RFx,x] && IGtQ[n,0]
```

2: $\int RF_x\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+$, then

$$\int RF_x\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx\;\longrightarrow\;\int \text{ExpandIntegrand}[RF_x\,(a+b\,\text{ArcSinh}[c\,x])^n,\,x]\,dx$$

Program code:

```
Int[RFx_*(a_+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[RFx*(a+b*ArcSinh[c*x])^n,x],x] /;
FreeQ[{a,b,c},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0]
```

```
Int[RFx_*(a_+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[RFx*(a+b*ArcCosh[c*x])^n,x],x] /;
FreeQ[{a,b,c},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0]
```

2. $\int RF_x\,\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+\bigwedge e==c^2\,d\bigwedge p-\frac{1}{2}\in\mathbb{Z}$

1: $\int RF_x\,\left(d+e\,x^2\right)^p\,\text{ArcSinh}[c\,x]^n\,dx$ when $n\in\mathbb{Z}^+\bigwedge e==c^2\,d\bigwedge p-\frac{1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+\bigwedge e==c^2\,d\bigwedge p-\frac{1}{2}\in\mathbb{Z}$, then

$$\int RF_x\,\left(d+e\,x^2\right)^p\,\text{ArcSinh}[c\,x]^n\,dx\;\longrightarrow\;\int \left(d+e\,x^2\right)^p\,\text{ArcSinh}[c\,x]^n\,\text{ExpandIntegrand}[RF_x,\,x]\,dx$$

Program code:

```
Int[RFx_*(d_+e_.*x_^2)^p_*ArcSinh[c_.*x_]^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[(d+e*x^2)^p*ArcSinh[c*x]^n,RFx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{c,d,e},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0] && EqQ[e,c^2*d] && IntegerQ[p-1/2]
```

```
Int[RFx_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*ArcCosh[c_.*x_]^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[(d1+e1*x)^p*(d2+e2*x)^p*ArcCosh[c*x]^n,RFx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{c,d1,e1,d2,e2},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[p-1/2]
```

2: $\int RF_x \left(d + e\,x^2\right)^p (a + b\,\text{ArcSinh}[c\,x])^n \, dx$ when $n \in \mathbb{Z}^+ \bigwedge e == c^2 d \bigwedge p - \frac{1}{2} \in \mathbb{Z}$

- **Derivation: Algebraic expansion**
- **Rule: If** $n \in \mathbb{Z}^+ \bigwedge e == c^2 d \bigwedge p - \frac{1}{2} \in \mathbb{Z}$, **then**

$$\int RF_x \left(d + e\,x^2\right)^p (a + b\,\text{ArcSinh}[c\,x])^n \, dx \longrightarrow \int \left(d + e\,x^2\right)^p \text{ExpandIntegrand}[RF_x\,(a + b\,\text{ArcSinh}[c\,x])^n,\,x] \, dx$$

- **Program code:**

```
Int[RFx_*(d_+e_.*x_^2)^p_*(a_+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)^p,RFx*(a+b*ArcSinh[c*x])^n,x],x] /;
FreeQ[{a,b,c,d,e},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0] && EqQ[e,c^2*d] && IntegerQ[p-1/2]
```

```
Int[RFx_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(d1+e1*x)^p*(d2+e2*x)^p,RFx*(a+b*ArcCosh[c*x])^n,x],x] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[p-1/2]
```

X: $\int u\,(a + b\,\text{ArcSinh}[c\,x])^n \, dx$

- **Rule:**

$$\int u\,(a + b\,\text{ArcSinh}[c\,x])^n \, dx \longrightarrow \int u\,(a + b\,\text{ArcSinh}[c\,x])^n \, dx$$

- **Program code:**

```
Int[u_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[u*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,n},x]
```

```
Int[u_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[u*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,n},x]
```