Rules for integrands of the form $\text{Trig}[d+e\,x]^m\left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p$

1. $\int\left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p\,dx$

   1. $\int\left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c=0$

      1: $\int\left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c=0\,\land\,p\in\mathbb{Z}$

- Derivation: Algebraic simplification
- Basis: If $b^2-4\,a\,c=0$, then $a+b\,z+c\,z^2=\frac{(b+2\,c\,z)^2}{4\,c}$
- Rule: If $b^2-4\,a\,c=0\,\land\,p\in\mathbb{Z}$, then

$$\int\left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p\,dx\,\longrightarrow\,\frac{1}{4^p\,c^p}\int(b+2\,c\,\text{Sec}[d+e\,x]^n)^{2\,p}\,dx$$

- Program code:

```
Int[(a_.+b_.*sec[d_.+e_.*x_]^n_.+c_.*sec[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  1/(4^p*c^p)*Int[(b+2*c*Sec[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

```
Int[(a_.+b_.*csc[d_.+e_.*x_]^n_.+c_.*csc[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  1/(4^p*c^p)*Int[(b+2*c*Csc[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

2: $\int \left(a + b \operatorname{Sec}[d + e x]^n + c \operatorname{Sec}[d + e x]^{2n}\right)^p dx$ when $b^2 - 4 a c == 0 \land p \notin \mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $b^2 - 4 a c == 0$**, then** $\partial_x \frac{(a+b F[x]+c F[x]^2)^p}{(b+2 c F[x])^{2p}} == 0$
- **Rule: If** $b^2 - 4 a c == 0 \land p \notin \mathbb{Z}$**, then**

$$\int \left(a + b \operatorname{Sec}[d + e x]^n + c \operatorname{Sec}[d + e x]^{2n}\right)^p dx \longrightarrow \frac{\left(a + b \operatorname{Sec}[d + e x]^n + c \operatorname{Sec}[d + e x]^{2n}\right)^p}{(b + 2 c \operatorname{Sec}[d + e x]^n)^{2p}} \int (b + 2 c \operatorname{Sec}[d + e x]^n)^{2p} dx$$

- **Program code:**

```
Int[(a_.+b_.*sec[d_.+e_.*x_]^n_.+c_.*sec[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Sec[d+e*x]^n+c*Sec[d+e*x]^(2*n))^p/(b+2*c*Sec[d+e*x]^n)^(2*p)*Int[u*(b+2*c*Sec[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

```
Int[(a_.+b_.*csc[d_.+e_.*x_]^n_.+c_.*csc[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Csc[d+e*x]^n+c*Csc[d+e*x]^(2*n))^p/(b+2*c*Csc[d+e*x]^n)^(2*p)*Int[u*(b+2*c*Csc[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

2. $\int \left(a + b \operatorname{Sec}[d + e x]^n + c \operatorname{Sec}[d + e x]^{2 n}\right)^p \, dx$ when $b^2 - 4 a c \neq 0$

1: $\int \frac{1}{a + b \operatorname{Sec}[d + e x]^n + c \operatorname{Sec}[d + e x]^{2 n}} \, dx$ when $b^2 - 4 a c \neq 0$

Derivation: Algebraic expansion

- **Basis: If** $q = \sqrt{b^2 - 4 a c}$ **, then** $\frac{1}{a+b z+c z^2} == \frac{2 c}{q (b-q+2 c z)} - \frac{2 c}{q (b+q+2 c z)}$

- **Rule: If** $b^2 - 4 a c \neq 0$**, let** $q = \sqrt{b^2 - 4 a c}$ **, then**

$$\int \frac{1}{a + b \operatorname{Sec}[d + e x]^n + c \operatorname{Sec}[d + e x]^{2 n}} \, dx \longrightarrow \frac{2 c}{q} \int \frac{1}{b - q + 2 c \operatorname{Sec}[d + e x]^n} \, dx - \frac{2 c}{q} \int \frac{1}{b + q + 2 c \operatorname{Sec}[d + e x]^n} \, dx$$

Program code:

```
Int[1/(a_.+b_.*sec[d_.+e_.*x_]^n_.+c_.*sec[d_.+e_.*x_]^n2_.),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  2*c/q*Int[1/(b-q+2*c*Sec[d+e*x]^n),x] -
  2*c/q*Int[1/(b+q+2*c*Sec[d+e*x]^n),x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0]
```

```
Int[1/(a_.+b_.*csc[d_.+e_.*x_]^n_.+c_.*csc[d_.+e_.*x_]^n2_.),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  2*c/q*Int[1/(b-q+2*c*Csc[d+e*x]^n),x] -
  2*c/q*Int[1/(b+q+2*c*Csc[d+e*x]^n),x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0]
```

2. $\int \operatorname{Sin}[d + e x]^m \left(a + b \operatorname{Sec}[d + e x]^n + c \operatorname{Sec}[d + e x]^{2 n}\right)^p \, dx$

1: $\int \operatorname{Sin}[d + e x]^m \left(a + b \operatorname{Sec}[d + e x]^n + c \operatorname{Sec}[d + e x]^{2 n}\right)^p \, dx$ when $\frac{m-1}{2} \in \mathbb{Z} \bigwedge n \in \mathbb{Z} \bigwedge p \in \mathbb{Z}$

Derivation: Integration by substitution

- **Basis: If** $\frac{m-1}{2} \in \mathbb{Z}$**, then** $\operatorname{Sin}[d + e x]^m F[\operatorname{Sec}[d + e x]] == -\frac{1}{e} \operatorname{Subst}\left[\left(1 - x^2\right)^{\frac{m-1}{2}} F\left[\frac{1}{x}\right], x, \operatorname{Cos}[d + e x]\right] \partial_x \operatorname{Cos}[d + e x]$

- **Rule: If** $\frac{m-1}{2} \in \mathbb{Z} \bigwedge n \in \mathbb{Z} \bigwedge p \in \mathbb{Z}$**, then**

$$\int \text{Sin}[d+e\,x]^m\left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p dx \to -\frac{1}{e}\,\text{Subst}\left[\int \frac{\left(1-x^2\right)^{\frac{m-1}{2}}\left(c+b\,x^n+a\,x^{2\,n}\right)^p}{x^{2\,n\,p}}\,dx,\,x,\,\text{Cos}[d+e\,x]\right]$$

Program code:

```
Int[sin[d_.+e_.*x_]^m_.*(a_.+b_.*sec[d_.+e_.*x_]^n_.+c_.*sec[d_.+e_.*x_]^n2_)^p_.,x_Symbol] :=
  Module[{f=FreeFactors[Cos[d+e*x],x]},
  -f/e*Subst[Int[(1-f^2*x^2)^((m-1)/2)*(b+a*(f*x)^n)^p/(f*x)^(n*p),x],x,Cos[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[(m-1)/2] && IntegersQ[n,p]
```

```
Int[cos[d_.+e_.*x_]^m_.*(a_.+b_.*csc[d_.+e_.*x_]^n_.+c_.*csc[d_.+e_.*x_]^n2_)^p_.,x_Symbol] :=
  Module[{f=FreeFactors[Sin[d+e*x],x]},
  f/e*Subst[Int[(1-f^2*x^2)^((m-1)/2)*(b+a*(f*x)^n)^p/(f*x)^(n*p),x],x,Sin[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[(m-1)/2] && IntegersQ[n,p]
```

2: $\int \text{Sin}[d+e\,x]^m\left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m}{2}\in\mathbb{Z}\bigwedge\frac{n}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

- **Basis:** $\text{Sin}[z]^2 == \frac{\text{Tan}[z]^2}{1+\text{Tan}[z]^2}$
- **Basis:** $\text{Sec}[z]^2 == 1+\text{Tan}[z]^2$
- **Basis: If** $\frac{m}{2}\in\mathbb{Z}$, **then** $\text{Sin}[d+e\,x]^m\,F\left[\text{Sec}[d+e\,x]^2\right] == \frac{1}{e}\,\text{Subst}\left[\frac{x^m\,F\left[1+x^2\right]}{\left(1+x^2\right)^{m/2+1}},\,x,\,\text{Tan}[d+e\,x]\right]\partial_x\text{Tan}[d+e\,x]$
- **Rule: If** $\frac{m}{2}\in\mathbb{Z}\bigwedge\frac{n}{2}\in\mathbb{Z}$, **then**

$$\int \text{Sin}[d+e\,x]^m\left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p dx \to \frac{1}{e}\,\text{Subst}\left[\int \frac{x^m\left(a+b\left(1+x^2\right)^{n/2}+c\left(1+x^2\right)^n\right)^p}{\left(1+x^2\right)^{m/2+1}}\,dx,\,x,\,\text{Tan}[d+e\,x]\right]$$

Program code:

```
Int[sin[d_.+e_.*x_]^m_*(a_.+b_.*sec[d_.+e_.*x_]^n_+c_.*sec[d_.+e_.*x_]^n2_)^p_.,x_Symbol] :=
  Module[{f=FreeFactors[Tan[d+e*x],x]},
  f^(m+1)/e*Subst[Int[x^m*ExpandToSum[a+b*(1+f^2*x^2)^(n/2)+c*(1+f^2*x^2)^n,x]^p/(1+f^2*x^2)^(m/2+1),x],x,Tan[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[n2,2*n] && IntegerQ[m/2] && IntegerQ[n/2]
```

```
Int[cos[d_.+e_.*x_]^m_*(a_.+b_.*csc[d_.+e_.*x_]^n_+c_.*csc[d_.+e_.*x_]^n2_)^p_.,x_Symbol] :=
  Module[{f=FreeFactors[Cot[d+e*x],x]},
  -f^(m+1)/e*Subst[Int[x^m*ExpandToSum[a+b*(1+f^2*x^2)^(n/2)+c*(1+f^2*x^2)^n,x]^p/(1+f^2*x^2)^(m/2+1),x],x,Cot[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[n2,2*n] && IntegerQ[m/2] && IntegerQ[n/2]
```

3. $\int \mathrm{Sec}[d+e\,x]^m \left(a+b\,\mathrm{Sec}[d+e\,x]^n+c\,\mathrm{Sec}[d+e\,x]^{2\,n}\right)^p\,dx$

1. $\int \mathrm{Sec}[d+e\,x]^m \left(a+b\,\mathrm{Sec}[d+e\,x]^n+c\,\mathrm{Sec}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0$

1: $\int \mathrm{Sec}[d+e\,x]^m \left(a+b\,\mathrm{Sec}[d+e\,x]^n+c\,\mathrm{Sec}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0\,\wedge\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

- **Basis: If** $b^2-4\,a\,c==0$, **then** $a+b\,z+c\,z^2==\frac{(b+2\,c\,z)^2}{4\,c}$
- **Rule: If** $b^2-4\,a\,c==0\,\wedge\,p\in\mathbb{Z}$, **then**

$$\int \mathrm{Sec}[d+e\,x]^m \left(a+b\,\mathrm{Sec}[d+e\,x]^n+c\,\mathrm{Sec}[d+e\,x]^{2\,n}\right)^p\,dx\;\longrightarrow\;\frac{1}{4^p\,c^p}\int \mathrm{Sec}[d+e\,x]^m\left(b+2\,c\,\mathrm{Sec}[d+e\,x]^n\right)^{2\,p}\,dx$$

Program code:

```
Int[sec[d_.+e_.*x_]^m_.*(a_.+b_.*sec[d_.+e_.*x_]^n_.+c_.*sec[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  1/(4^p*c^p)*Int[Sec[d+e*x]^m*(b+2*c*Sec[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

```
Int[csc[d_.+e_.*x_]^m_.*(a_.+b_.*csc[d_.+e_.*x_]^n_.+c_.*csc[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  1/(4^p*c^p)*Int[Csc[d+e*x]^m*(b+2*c*Csc[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

2: $\int \text{Sec}[d+e\,x]^m \left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p dx$ when $b^2-4\,a\,c==0\ \wedge\ p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c==0$**, then** $\partial_x \frac{(a+b\,F[x]+c\,F[x]^2)^p}{(b+2\,c\,F[x])^{2\,p}}==0$

Rule: If $b^2-4\,a\,c==0\ \wedge\ p\notin\mathbb{Z}$**, then**

$$\int \text{Sec}[d+e\,x]^m \left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p dx \longrightarrow \frac{\left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p}{(b+2\,c\,\text{Sec}[d+e\,x]^n)^{2\,p}}\int \text{Sec}[d+e\,x]^m\,(b+2\,c\,\text{Sec}[d+e\,x]^n)^{2\,p}\,dx$$

Program code:

```
Int[sec[d_.+e_.*x_]^m_.*(a_.+b_.*sec[d_.+e_.*x_]^n_.+c_.*sec[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Sec[d+e*x]^n+c*Sec[d+e*x]^(2*n))^p/(b+2*c*Sec[d+e*x]^n)^(2*p)*Int[Sec[d+e*x]^m*(b+2*c*Sec[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

```
Int[csc[d_.+e_.*x_]^m_.*(a_.+b_.*csc[d_.+e_.*x_]^n_.+c_.*csc[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Csc[d+e*x]^n+c*Csc[d+e*x]^(2*n))^p/(b+2*c*Csc[d+e*x]^n)^(2*p)*Int[Csc[d+e*x]^m*(b+2*c*Csc[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

2. $\int \text{Sec}[d+e\,x]^m \left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0$

1: $\int \text{Sec}[d+e\,x]^m \left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p dx$ when $(m\mid n\mid p)\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $(m\mid n\mid p)\in\mathbb{Z}$**, then**

$$\int \text{Sec}[d+e\,x]^m \left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p dx \longrightarrow \int \text{ExpandTrig}\left[\text{Sec}[d+e\,x]^m \left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p,\ x\right] dx$$

Proeram code:

```
Int[sec[d_.+e_.*x_]^m_.*(a_.+b_.*sec[d_.+e_.*x_]^n_.+c_.*sec[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  Int[ExpandTrig[sec[d+e*x]^m*(a+b*sec[d+e*x]^n+c*sec[d+e*x]^(2*n))^p,x],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegersQ[m,n,p]
```

```
Int[csc[d_.+e_.*x_]^m_.*(a_.+b_.*csc[d_.+e_.*x_]^n_.+c_.*csc[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  Int[ExpandTrig[csc[d+e*x]^m*(a+b*csc[d+e*x]^n+c*csc[d+e*x]^(2*n))^p,x],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegersQ[m,n,p]
```

4. $\int \text{Tan}[d+e\,x]^m \left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p \,dx$

1: $\int \text{Tan}[d+e\,x]^m \left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p \,dx$ when $\frac{m-1}{2} \in \mathbb{Z} \bigwedge n \in \mathbb{Z} \bigwedge p \in \mathbb{Z}$

Derivation: Integration by substitution

- **Basis:** $\text{Tan}[z]^2 == \frac{1-\text{Cos}[z]^2}{\text{Cos}[z]^2}$

- **Basis: If** $\frac{m-1}{2} \in \mathbb{Z}$, **then** $\text{Tan}[d+e\,x]^m\,F[\text{Sec}[d+e\,x]] == -\frac{1}{e}\,\text{Subst}\left[\frac{(1-x^2)^{\frac{m-1}{2}}\,F\left[\frac{1}{x}\right]}{x^m},\,x,\,\text{Cos}[d+e\,x]\right]\,\partial_x\text{Cos}[d+e\,x]$

- **Rule: If** $\frac{m-1}{2} \in \mathbb{Z} \bigwedge n \in \mathbb{Z} \bigwedge p \in \mathbb{Z}$, **then**

$$\int \text{Tan}[d+e\,x]^m \left(a+b\,\text{Sec}[d+e\,x]^n+c\,\text{Sec}[d+e\,x]^{2\,n}\right)^p dx \longrightarrow -\frac{1}{e}\,\text{Subst}\left[\int \frac{\left(1-x^2\right)^{\frac{m-1}{2}}\left(c+b\,x^n+a\,x^{2\,n}\right)^p}{x^{m+2\,n\,p}}\,dx,\,x,\,\text{Cos}[d+e\,x]\right]$$

Program code:

```
Int[tan[d_.+e_.*x_]^m_.*(a_+b_.*sec[d_.+e_.*x_]^n_.+c_.*sec[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  Module[{f=FreeFactors[Cos[d+e*x],x]},
  -1/(e*f^(m+n*p-1))*Subst[Int[(1-f^2*x^2)^((m-1)/2)*(c+b*(f*x)^n+c*(f*x)^(2*n))^p/x^(m+2*n*p),x],x,Cos[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && IntegerQ[(m-1)/2] && IntegerQ[n] && IntegerQ[p]
```

```
Int[cot[d_.+e_.*x_]^m_.*(a_+b_.*csc[d_.+e_.*x_]^n_.+c_.*sec[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  Module[{f=FreeFactors[Sin[d+e*x],x]},
  1/(e*f^(m+n*p-1))*Subst[Int[(1-f^2*x^2)^((m-1)/2)*(c+b*(f*x)^n+c*(f*x)^(2*n))^p/x^(m+2*n*p),x],x,Sin[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && IntegerQ[(m-1)/2] && IntegerQ[n] && IntegerQ[p]
```

2: $\int \operatorname{Tan}[d+e\,x]^m \left(a+b\operatorname{Sec}[d+e\,x]^n+c\operatorname{Sec}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m}{2}\in\mathbb{Z}\bigwedge\frac{n}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

- **Basis:** $\operatorname{Sec}[z]^2 == 1+\operatorname{Tan}[z]^2$

- **Basis:** $\operatorname{Tan}[d+e\,x]^m F\left[\operatorname{Sec}[d+e\,x]^2\right] == \frac{1}{e}\operatorname{Subst}\left[\frac{x^m F[1+x^2]}{1+x^2},\ x,\ \operatorname{Tan}[d+e\,x]\right]\partial_x\operatorname{Tan}[d+e\,x]$

- **Rule:** If $\frac{m}{2}\in\mathbb{Z}\bigwedge\frac{n}{2}\in\mathbb{Z}$, then

$$\int \operatorname{Tan}[d+e\,x]^m \left(a+b\operatorname{Sec}[d+e\,x]^n+c\operatorname{Sec}[d+e\,x]^{2\,n}\right)^p dx \rightarrow \frac{1}{e}\operatorname{Subst}\left[\int\frac{x^m\left(a+b\left(1+x^2\right)^{n/2}+c\left(1+x^2\right)^n\right)^p}{1+x^2}dx,\ x,\ \operatorname{Tan}[d+e\,x]\right]$$

Program code:

```
Int[tan[d_.+e_.*x_]^m_.*(a_+b_.*sec[d_.+e_.*x_]^n_+c_.*sec[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  Module[{f=FreeFactors[Tan[d+e*x],x]},
  f^(m+1)/e*Subst[Int[x^m*ExpandToSum[a+b*(1+f^2*x^2)^(n/2)+c*(1+f^2*x^2)^n,x]^p/(1+f^2*x^2),x],x,Tan[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[m/2] && IntegerQ[n/2]
```

```
Int[cot[d_.+e_.*x_]^m_.*(a_+b_.*csc[d_.+e_.*x_]^n_+c_.*sec[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  Module[{f=FreeFactors[Cot[d+e*x],x]},
  -f^(m+1)/e*Subst[Int[x^m*ExpandToSum[a+b*(1+f^2*x^2)^(n/2)+c*(1+f^2*x^2)^n,x]^p/(1+f^2*x^2),x],x,Cot[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[m/2] && IntegerQ[n/2]
```

5. $\int (A + B\,\text{Sec}[d + e\,x])\,\left(a + b\,\text{Sec}[d + e\,x] + c\,\text{Sec}[d + e\,x]^2\right)^n\,dx$

1. $\int (A + B\,\text{Sec}[d + e\,x])\,\left(a + b\,\text{Sec}[d + e\,x] + c\,\text{Sec}[d + e\,x]^2\right)^n\,dx$ when $b^2 - 4\,a\,c == 0$

1: $\int (A + B\,\text{Sec}[d + e\,x])\,\left(a + b\,\text{Sec}[d + e\,x] + c\,\text{Sec}[d + e\,x]^2\right)^n\,dx$ when $b^2 - 4\,a\,c == 0\,\land\,n \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2 - 4\,a\,c == 0$, then $a + b\,z + c\,z^2 == \frac{(b+2\,c\,z)^2}{4\,c}$

Rule: If $b^2 - 4\,a\,c == 0\,\land\,n \in \mathbb{Z}$, then

$$\int (A + B\,\text{Sec}[d + e\,x])\,\left(a + b\,\text{Sec}[d + e\,x] + c\,\text{Sec}[d + e\,x]^2\right)^n\,dx \,\longrightarrow\, \frac{1}{4^n\,c^n}\int (A + B\,\text{Sec}[d + e\,x])\,(b + 2\,c\,\text{Sec}[d + e\,x])^{2\,n}\,dx$$

Program code:

```
Int[(A_+B_.*sec[d_.+e_.*x_])*(a_+b_.*sec[d_.+e_.*x_]+c_.*sec[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  1/(4^n*c^n)*Int[(A+B*Sec[d+e*x])*(b+2*c*Sec[d+e*x])^(2*n),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && EqQ[b^2-4*a*c,0] && IntegerQ[n]
```

```
Int[(A_+B_.*csc[d_.+e_.*x_])*(a_+b_.*csc[d_.+e_.*x_]+c_.*csc[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  1/(4^n*c^n)*Int[(A+B*Csc[d+e*x])*(b+2*c*Csc[d+e*x])^(2*n),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && EqQ[b^2-4*a*c,0] && IntegerQ[n]
```

2: $\int (A + B \operatorname{Sec}[d + e x]) \left(a + b \operatorname{Sec}[d + e x] + c \operatorname{Sec}[d + e x]^2\right)^n \, dx$ when $b^2 - 4 a c == 0 \wedge n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis: If** $b^2 - 4 a c == 0$, **then** $\partial_x \frac{(a+b F[x]+c F[x]^2)^n}{(b+2 c F[x])^{2 n}} == 0$

- **Rule: If** $b^2 - 4 a c == 0 \wedge n \notin \mathbb{Z}$, **then**

$$\int (A + B \operatorname{Sec}[d + e x]) \left(a + b \operatorname{Sec}[d + e x] + c \operatorname{Sec}[d + e x]^2\right)^n \, dx \longrightarrow \frac{\left(a + b \operatorname{Sec}[d + e x] + c \operatorname{Sec}[d + e x]^2\right)^n}{(b + 2 c \operatorname{Sec}[d + e x])^{2 n}} \int (A + B \operatorname{Sec}[d + e x]) \, (b + 2 c \operatorname{Sec}[d + e x])^{2 n} \, dx$$

Program code:

```
Int[(A_+B_.*sec[d_.+e_.*x_])*(a_+b_.*sec[d_.+e_.*x_]+c_.*sec[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  (a+b*Sec[d+e*x]+c*Sec[d+e*x]^2)^n/(b+2*c*Sec[d+e*x])^(2*n)*Int[(A+B*Sec[d+e*x])*(b+2*c*Sec[d+e*x])^(2*n),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[n]]
```

```
Int[(A_+B_.*csc[d_.+e_.*x_])*(a_+b_.*csc[d_.+e_.*x_]+c_.*csc[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  (a+b*Csc[d+e*x]+c*Csc[d+e*x]^2)^n/(b+2*c*Csc[d+e*x])^(2*n)*Int[(A+B*Csc[d+e*x])*(b+2*c*Csc[d+e*x])^(2*n),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[n]]
```

2. $\int (A + B \operatorname{Sec}[d + e x]) \left(a + b \operatorname{Sec}[d + e x] + c \operatorname{Sec}[d + e x]^2\right)^n dx$ when $b^2 - 4 a c \neq 0$

1: $\int \frac{A + B \operatorname{Sec}[d + e x]}{a + b \operatorname{Sec}[d + e x] + c \operatorname{Sec}[d + e x]^2} dx$ when $b^2 - 4 a c \neq 0$

Derivation: Algebraic expansion

- **Basis: If** $q = \sqrt{b^2 - 4 a c}$, **then** $\frac{A+B z}{a+b z+c z^2} == \left(B + \frac{b B-2 A c}{q}\right) \frac{1}{b+q+2 c z} + \left(B - \frac{b B-2 A c}{q}\right) \frac{1}{b-q+2 c z}$

- **Rule: If** $b^2 - 4 a c \neq 0$, **let** $q = \sqrt{b^2 - 4 a c}$, **then**

$$\int \frac{A + B \operatorname{Sec}[d + e x]}{a + b \operatorname{Sec}[d + e x] + c \operatorname{Sec}[d + e x]^2} dx \longrightarrow \left(B + \frac{b B - 2 A c}{q}\right) \int \frac{1}{b + q + 2 c \operatorname{Sec}[d + e x]} dx + \left(B - \frac{b B - 2 A c}{q}\right) \int \frac{1}{b - q + 2 c \operatorname{Sec}[d + e x]} dx$$

- **Program code:**

```
Int[(A_+B_.*sec[d_.+e_.*x_])/(a_.+b_.*sec[d_.+e_.*x_]+c_.*sec[d_.+e_.*x_]^2),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  (B+(b*B-2*A*c)/q)*Int[1/(b+q+2*c*Sec[d+e*x]),x] +
  (B-(b*B-2*A*c)/q)*Int[1/(b-q+2*c*Sec[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0]
```

```
Int[(A_+B_.*csc[d_.+e_.*x_])/(a_.+b_.*csc[d_.+e_.*x_]+c_.*csc[d_.+e_.*x_]^2),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  (B+(b*B-2*A*c)/q)*Int[1/(b+q+2*c*Csc[d+e*x]),x] +
  (B-(b*B-2*A*c)/q)*Int[1/(b-q+2*c*Csc[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0]
```

2: $\int (A+B\,\mathrm{Sec}[d+e\,x])\left(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Sec}[d+e\,x]^2\right)^n dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}$

- **Derivation: Algebraic expansion**
- **Rule: If** $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}$

$$\int (A+B\,\mathrm{Sec}[d+e\,x])\left(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Sec}[d+e\,x]^2\right)^n dx \longrightarrow \int \mathrm{ExpandTrig}\left[(A+B\,\mathrm{Sec}[d+e\,x])\left(a+b\,\mathrm{Sec}[d+e\,x]+c\,\mathrm{Sec}[d+e\,x]^2\right)^n,\,x\right] dx$$

- **Program code:**

```
Int[(A_+B_.*sec[d_.+e_.*x_])*(a_.+b_.*sec[d_.+e_.*x_]+c_.*sec[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  Int[ExpandTrig[(A+B*sec[d+e*x])*(a+b*sec[d+e*x]+c*sec[d+e*x]^2)^n,x],x] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0] && IntegerQ[n]
```

```
Int[(A_+B_.*csc[d_.+e_.*x_])*(a_.+b_.*csc[d_.+e_.*x_]+c_.*csc[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  Int[ExpandTrig[(A+B*csc[d+e*x])*(a+b*csc[d+e*x]+c*csc[d+e*x]^2)^n,x],x] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0] && IntegerQ[n]
```