Rules for integrands of the form $(d\,x)^m\,\left(a+b\,x^2+c\,x^4\right)^p$

x. $\int (d\,x)^m\,\left(b\,x^2+c\,x^4\right)^p\,dx$

1: $\int (d\,x)^m\,\left(b\,x^2+c\,x^4\right)^p\,dx$ when $p\in\mathbb{Z}$

- **Derivation: Algebraic simplification**
- **Basis: If $p\in\mathbb{Z}$, then** $\left(b\,x^2+c\,x^4\right)^p==\frac{1}{d^{2\,p}}\,(d\,x)^{2\,p}\,\left(b+c\,x^2\right)^p$
- **Rule 1.2.2.2.0.1: If $p\in\mathbb{Z}$, then**

$$\int (d\,x)^m\,\left(b\,x^2+c\,x^4\right)^p\,dx\;\longrightarrow\;\frac{1}{d^{2\,p}}\int (d\,x)^{m+2\,p}\,\left(b+c\,x^2\right)^p\,dx$$

- **Program code:**

```
(* Int[(d_.*x_)^m_.*(b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  1/d^(2*p)*Int[(d*x)^(m+2*p)*(b+c*x^2)^p,x] /;
FreeQ[{b,c,d,m},x] && IntegerQ[p] *)
```

2: $\int (d\,x)^m\,\left(b\,x^2+c\,x^4\right)^p\,dx$ when $p\notin\mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x\frac{\left(b\,x^2+c\,x^4\right)^p}{(d\,x)^{2\,p}\,\left(b+c\,x^2\right)^p}==0$
- **Rule 1.2.2.2.0.2: If $p\notin\mathbb{Z}$, then**

$$\int (d\,x)^m\,\left(b\,x^2+c\,x^4\right)^p\,dx\;\longrightarrow\;\frac{\left(b\,x^2+c\,x^4\right)^p}{(d\,x)^{2\,p}\,\left(b+c\,x^2\right)^p}\int (d\,x)^{m+2\,p}\,\left(b+c\,x^2\right)^p\,dx$$

- **Program code:**

```
(* Int[(d_.*x_)^m_.*(b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (b*x^2+c*x^4)^p/((d*x)^(2*p)*(b+c*x^2)^p)*Int[(d*x)^(m+2*p)*(b+c*x^2)^p,x] /;
FreeQ[{b,c,d,m,p},x] && Not[IntegerQ[p]] *)
```

1: $\int x\left(a+b x^2+c x^4\right)^p dx$

- **Derivation: Integration by substitution**
- **Basis:** $x F\left[x^2\right] == \frac{1}{2} \text{Subst}\left[F[x],\ x,\ x^2\right] \partial_x x^2$

Rule 1.2.2.2.1:

$$\int x\left(a+b x^2+c x^4\right)^p dx \longrightarrow \frac{1}{2} \text{Subst}\left[\int\left(a+b x+c x^2\right)^p dx,\ x,\ x^2\right]$$

Program code:

```
Int[x_*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  1/2*Subst[Int[(a+b*x+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,b,c,p},x]
```

2: $\int (d x)^m\left(a+b x^2+c x^4\right)^p dx$ when $p \in \mathbb{Z}^+$

- **Derivation: Algebraic expansion**
- **Rule 1.2.2.2.2: If** $p \in \mathbb{Z}^+$, **then**

$$\int (d x)^m\left(a+b x^2+c x^4\right)^p dx \longrightarrow \int \text{ExpandIntegrand}\left[(d x)^m\left(a+b x^2+c x^4\right)^p,\ x\right] dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d*x)^m*(a+b*x^2+c*x^4)^p,x],x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[p,0] && Not[IntegerQ[(m+1)/2]]
```

3. $\int (d x)^m\left(a+b x^2+c x^4\right)^p dx$ when $b^2-4 a c == 0$

x: $\int (d x)^m\left(a+b x^2+c x^4\right)^p dx$ when $b^2-4 a c == 0 \land p \in \mathbb{Z}$

- **Derivation: Algebraic simplification**
- **Basis: If** $b^2-4 a c == 0$, **then** $a+b z+c z^2 == \frac{1}{c}\left(\frac{b}{2}+c z\right)^2$
- **Rule 1.2.2.2.3.1: If** $b^2-4 a c == 0 \land p \in \mathbb{Z}$, **then**

$$\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx \longrightarrow \frac{1}{c^p}\int (d\,x)^m \left(\frac{b}{2}+c\,x^2\right)^{2\,p} dx$$

Program code:

```
(* Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  1/c^p*Int[(d*x)^m*(b/2+c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p] *)
```

2. $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx$ when $b^2-4\,a\,c=0 \wedge p\notin\mathbb{Z}$

 1. $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx$ when $b^2-4\,a\,c=0 \wedge p\notin\mathbb{Z} \wedge m+4\,p+5=0 \wedge p\neq -\frac{1}{2}$

  1: $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx$ when $b^2-4\,a\,c=0 \wedge p\notin\mathbb{Z} \wedge m+4\,p+5=0 \wedge p<-1$

Derivation: Square trinomial recurrence 2c with $m+4\,p+5=0$

Rule 1.2.2.2.3.2.1: If $b^2-4\,a\,c=0 \wedge p\notin\mathbb{Z} \wedge m+4\,p+5=0 \wedge p<-1$, then

$$\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx \longrightarrow \frac{2\,(d\,x)^{m+1}\left(a+b\,x^2+c\,x^4\right)^{p+1}}{d\,(m+3)\left(2\,a+b\,x^2\right)} - \frac{(d\,x)^{m+1}\left(a+b\,x^2+c\,x^4\right)^{p+1}}{2\,a\,d\,(m+3)\,(p+1)}$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  2*(d*x)^(m+1)*(a+b*x^2+c*x^4)^(p+1)/(d*(m+3)*(2*a+b*x^2)) -
  (d*x)^(m+1)*(a+b*x^2+c*x^4)^(p+1)/(2*a*d*(m+3)*(p+1)) /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]] && EqQ[m+4*p+5,0] && LtQ[p,-1]
```

2: $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p\,dx$ when $b^2-4\,a\,c==0\;\bigwedge\;p\notin\mathbb{Z}\;\bigwedge\;m+4\,p+5==0\;\bigwedge\;p\neq-\frac{1}{2}$

Derivation: Square trinomial recurrence 2c with $m+4\,p+5==0$

Rule 1.2.2.2.3.2.1: If $b^2-4\,a\,c==0\;\bigwedge\;p\notin\mathbb{Z}\;\bigwedge\;m+4\,p+5==0\;\bigwedge\;p\neq-\frac{1}{2}$, then

$$\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p\,dx\;\longrightarrow\;\frac{(d\,x)^{m+1}\left(a+b\,x^2+c\,x^4\right)^{p+1}}{4\,a\,d\,(p+1)\,(2\,p+1)}-\frac{(d\,x)^{m+1}\left(2\,a+b\,x^2\right)\left(a+b\,x^2+c\,x^4\right)^p}{4\,a\,d\,(2\,p+1)}$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*x^2+c*x^4)^(p+1)/(4*a*d*(p+1)*(2*p+1)) -
  (d*x)^(m+1)*(2*a+b*x^2)*(a+b*x^2+c*x^4)^p/(4*a*d*(2*p+1)) /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]] && EqQ[m+4*p+5,0] && NeQ[p,-1/2]
```

?: $\int x^m \left(a+b\,x^2+c\,x^4\right)^p\,dx$ when $b^2-4\,a\,c==0\;\bigwedge\;p-\frac{1}{2}\in\mathbb{Z}\;\bigwedge\;\frac{m-1}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

- **Basis: If $\frac{m-1}{2}\in\mathbb{Z}$, then $x^m\,F\left[x^2\right]==\frac{1}{2}\,\text{Subst}\left[x^{\frac{m-1}{2}}\,F[x],\,x,\,x^2\right]\partial_x x^2$**

Rule 1.2.2.2.5.1: If $b^2-4\,a\,c==0\;\bigwedge\;p-\frac{1}{2}\in\mathbb{Z}\;\bigwedge\;\frac{m-1}{2}\in\mathbb{Z}$, then

$$\int x^m \left(a+b\,x^2+c\,x^4\right)^p\,dx\;\longrightarrow\;\frac{1}{2}\,\text{Subst}\left[\int x^{\frac{m-1}{2}}\left(a+b\,x+c\,x^2\right)^p\,dx,\,x,\,x^2\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  1/2*Subst[Int[x^((m-1)/2)*(a+b*x+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,b,c,p},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p-1/2] && IntegerQ[(m-1)/2] && (GtQ[m,0] || LtQ[0,4*p,-m-1])
```

2: $\int (d\,x)^m\,\left(a+b\,x^2+c\,x^4\right)^p\,dx$ when $b^2-4\,a\,c==0\,\bigwedge\,p-\frac{1}{2}\in\mathbb{Z}\,\bigwedge\,m\in\mathbb{Z}^+$ **Delete!**

Derivation: Piecewise constant extraction

- **Basis: If** $b^2-4\,a\,c==0$, **then** $\partial_x\frac{\left(a+b\,x^2+c\,x^4\right)^{p+1}}{\left(\frac{b}{2}+c\,x^2\right)^{2\,(p+1)}}==0$

- **Rule 1.2.2.2.3.2.2: If** $b^2-4\,a\,c==0\,\bigwedge\,p-\frac{1}{2}\in\mathbb{Z}\,\bigwedge\,m\in\mathbb{Z}^+$, **then**

$$\int (d\,x)^m\,\left(a+b\,x^2+c\,x^4\right)^p\,dx\;\longrightarrow\;\frac{c\,\left(a+b\,x^2+c\,x^4\right)^{p+1}}{\left(\frac{b}{2}+c\,x^2\right)^{2\,(p+1)}}\int (d\,x)^m\,\left(\frac{b}{2}+c\,x^2\right)^{2\,p}\,dx$$

Program code:

```
(* Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  c*(a+b*x^2+c*x^4)^(p+1)/(b/2+c*x^2)^(2*(p+1))*Int[(d*x)^m*(b/2+c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p-1/2] && IGeQ[m,2*p] *)
```

3: $\int (d\,x)^m\,\left(a+b\,x^2+c\,x^4\right)^p\,dx$ when $b^2-4\,a\,c==0\,\wedge\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis: If** $b^2-4\,a\,c==0$, **then** $\partial_x\frac{\left(a+b\,x^2+c\,x^4\right)^{p}}{\left(\frac{b}{2}+c\,x^2\right)^{2\,p}}==0$

- **Note: If** $b^2-4\,a\,c==0$, **then** $a+b\,z+c\,z^2==\frac{1}{c}\left(\frac{b}{2}+c\,z\right)^2$

- **Rule 1.2.2.2.3.2.2: If** $b^2-4\,a\,c==0\,\wedge\,p\notin\mathbb{Z}$, **then**

$$\int (d\,x)^m\,\left(a+b\,x^2+c\,x^4\right)^p\,dx\;\longrightarrow\;\frac{\left(a+b\,x^2+c\,x^4\right)^{\text{FracPart}[p]}}{c^{\text{IntPart}[p]}\,\left(\frac{b}{2}+c\,x^2\right)^{2\,\text{FracPart}[p]}}\int (d\,x)^m\,\left(\frac{b}{2}+c\,x^2\right)^{2\,p}\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (a+b*x^2+c*x^4)^FracPart[p]/(c^IntPart[p]*(b/2+c*x^2)^(2*FracPart[p]))*Int[(d*x)^m*(b/2+c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p-1/2]
```

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  a^IntPart[p]*(a+b*x^2+c*x^4)^FracPart[p]/(1+2*c*x^2/b)^(2*FracPart[p])*Int[(d*x)^m*(1+2*c*x^2/b)^(2*p),x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[2*p]]
```

4: $\int x^m \left(a+b x^2+c x^4\right)^p dx$ when $\frac{m-1}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

- **Basis: If** $\frac{m-1}{2} \in \mathbb{Z}$, **then** $x^m F\left[x^2\right] = \frac{1}{2} \text{Subst}\left[x^{\frac{m-1}{2}} F[x],\ x,\ x^2\right] \partial_x x^2$
- **Rule 1.2.2.2.5.1: If** $\frac{m-1}{2} \in \mathbb{Z}$, **then**

$$\int x^m \left(a+b x^2+c x^4\right)^p dx \longrightarrow \frac{1}{2} \text{Subst}\left[\int x^{\frac{m-1}{2}} \left(a+b x+c x^2\right)^p dx,\ x,\ x^2\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  1/2*Subst[Int[x^((m-1)/2)*(a+b*x+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,b,c,p},x] && IntegerQ[(m-1)/2]
```

5: $\int (d x)^m \left(a+b x^2+c x^4\right)^p dx$ when $b^2-4 a c \neq 0 \land m \in \mathbb{F}$

Derivation: Integration by substitution

- **Basis: If** $k \in \mathbb{Z}^+$, **then** $(d x)^m F[x] = \frac{k}{d} \text{Subst}\left[x^{k (m+1)-1} F\left[\frac{x^k}{d}\right],\ x,\ (d x)^{1/k}\right] \partial_x (d x)^{1/k}$
- **Rule 1.2.2.2.6.1.2: If** $b^2-4 a c \neq 0 \land m \in \mathbb{F}$, **let** $k = \text{Denominator}[m]$, **then**

$$\int (d x)^m \left(a+b x^2+c x^4\right)^p dx \longrightarrow \frac{k}{d} \text{Subst}\left[\int x^{k (m+1)-1} \left(a+\frac{b x^{2 k}}{d^2}+\frac{c x^{4 k}}{d^4}\right)^p dx,\ x,\ (d x)^{1/k}\right]$$

Program code:

```
Int[(d_.*x_)^m_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  With[{k=Denominator[m]},
  k/d*Subst[Int[x^(k*(m+1)-1)*(a+b*x^(2*k)/d^2+c*x^(4*k)/d^4)^p,x],x,(d*x)^(1/k)]] /;
FreeQ[{a,b,c,d,p},x] && NeQ[b^2-4*a*c,0] && FractionQ[m] && IntegerQ[p]
```

6. $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ p>0$

1: $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ p>0\ \wedge\ m>1$

Derivation: Trinomial recurrence 1b with $A=0$, $B=1$ and $m=m-n$

Rule 1.2.2.2.6.1.3.1: If $b^2-4\,a\,c\neq 0\ \wedge\ p>0\ \wedge\ m>1$, then

$$\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p\,dx\ \longrightarrow$$
$$\frac{d\,(d\,x)^{m-1}\left(a+b\,x^2+c\,x^4\right)^p\left(2\,b\,p+c\,(m+4\,p-1)\,x^2\right)}{c\,(m+4\,p+1)\,(m+4\,p-1)}-$$
$$\frac{2\,p\,d^2}{c\,(m+4\,p+1)\,(m+4\,p-1)}\int (d\,x)^{m-2}\left(a+b\,x^2+c\,x^4\right)^{p-1}\left(a\,b\,(m-1)-\left(2\,a\,c\,(m+4\,p-1)-b^2\,(m+2\,p-1)\right)x^2\right)dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  d*(d*x)^(m-1)*(a+b*x^2+c*x^4)^p*(2*b*p+c*(m+4*p-1)*x^2)/(c*(m+4*p+1)*(m+4*p-1)) -
  2*p*d^2/(c*(m+4*p+1)*(m+4*p-1))*
    Int[(d*x)^(m-2)*(a+b*x^2+c*x^4)^(p-1)*Simp[a*b*(m-1)-(2*a*c*(m+4*p-1)-b^2*(m+2*p-1))*x^2,x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && GtQ[p,0] && GtQ[m,1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

2: $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ p>0\ \wedge\ m<-1$

Reference: G&R 2.160.2

Derivation: Trinomial recurrence 1a with $A=1$ and $B=0$

Rule 1.2.2.2.6.1.3.2: If $b^2-4\,a\,c\neq 0\ \wedge\ p>0\ \wedge\ m<-1$, then

$$\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p\,dx\ \longrightarrow\ \frac{(d\,x)^{m+1}\left(a+b\,x^2+c\,x^4\right)^p}{d\,(m+1)}-\frac{2\,p}{d^2\,(m+1)}\int (d\,x)^{m+2}\left(b+2\,c\,x^2\right)\left(a+b\,x^2+c\,x^4\right)^{p-1}\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*x^2+c*x^4)^p/(d*(m+1)) -
  2*p/(d^2*(m+1))*Int[(d*x)^(m+2)*(b+2*c*x^2)*(a+b*x^2+c*x^4)^(p-1),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && GtQ[p,0] && LtQ[m,-1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

3: $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,p>0\,\wedge\,m+4\,p+1\neq 0$

Derivation: Trinomial recurrence 1a with $A=0$, $B=1$ and $m=m-n$

Derivation: Trinomial recurrence 1b with $A=1$ and $B=0$

Rule 1.2.2.2.6.1.3.4: If $b^2-4\,a\,c\neq 0\,\wedge\,p>0\,\wedge\,m+4\,p+1\neq 0$, then

$$\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx\;\longrightarrow\;\frac{(d\,x)^{m+1}\left(a+b\,x^2+c\,x^4\right)^p}{d\,(m+4\,p+1)}+\frac{2\,p}{m+4\,p+1}\int (d\,x)^m\left(2\,a+b\,x^2\right)\left(a+b\,x^2+c\,x^4\right)^{p-1}dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*x^2+c*x^4)^p/(d*(m+4*p+1)) +
  2*p/(m+4*p+1)*Int[(d*x)^m*(2*a+b*x^2)*(a+b*x^2+c*x^4)^(p-1),x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[b^2-4*a*c,0] && GtQ[p,0] && NeQ[m+4*p+1,0] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

7. $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx$ when $b^2-4\,a\,c\neq 0\ \land\ p<-1$

1. $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx$ when $b^2-4\,a\,c\neq 0\ \land\ p<-1\ \land\ m>1$

1: $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx$ when $b^2-4\,a\,c\neq 0\ \land\ p<-1\ \land\ 1<m\leq 3$

Derivation: Trinomial recurrence 2a with $A=1$ and $B=0$

Derivation: Trinomial recurrence 2b with $A=0$, $B=1$ and $m=m-n$

Rule 1.2.2.2.6.1.4.1.1: If $b^2-4\,a\,c\neq 0\ \land\ p<-1\ \land\ 1<m\leq 3$, then

$$\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx\ \longrightarrow$$

$$\frac{d\,(d\,x)^{m-1}\left(b+2\,c\,x^2\right)\left(a+b\,x^2+c\,x^4\right)^{p+1}}{2\,(p+1)\left(b^2-4\,a\,c\right)}-\frac{d^2}{2\,(p+1)\left(b^2-4\,a\,c\right)}\int (d\,x)^{m-2}\left(b\,(m-1)+2\,c\,(m+4\,p+5)\,x^2\right)\left(a+b\,x^2+c\,x^4\right)^{p+1}dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  d*(d*x)^(m-1)*(b+2*c*x^2)*(a+b*x^2+c*x^4)^(p+1)/(2*(p+1)*(b^2-4*a*c)) -
  d^2/(2*(p+1)*(b^2-4*a*c))*Int[(d*x)^(m-2)*(b*(m-1)+2*c*(m+4*p+5)*x^2)*(a+b*x^2+c*x^4)^(p+1),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && GtQ[m,1] && LeQ[m,3] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

2: $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,p<-1\,\wedge\,m>3$

Derivation: Trinomial recurrence 2a with $A=0$, $B=1$ and $m=m-n$

Rule 1.2.2.2.6.1.4.1.2: If $b^2-4\,a\,c\neq 0\,\wedge\,p<-1\,\wedge\,m>3$, then

$$\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx\;\longrightarrow$$

$$-\frac{d^3\,(d\,x)^{m-3}\left(2\,a+b\,x^2\right)\left(a+b\,x^2+c\,x^4\right)^{p+1}}{2\,(p+1)\left(b^2-4\,a\,c\right)}+\frac{d^4}{2\,(p+1)\left(b^2-4\,a\,c\right)}\int (d\,x)^{m-4}\left(2\,a\,(m-3)+b\,(m+4\,p+3)\,x^2\right)\left(a+b\,x^2+c\,x^4\right)^{p+1}dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  -d^3*(d*x)^(m-3)*(2*a+b*x^2)*(a+b*x^2+c*x^4)^(p+1)/(2*(p+1)*(b^2-4*a*c)) +
  d^4/(2*(p+1)*(b^2-4*a*c))*Int[(d*x)^(m-4)*(2*a*(m-3)+b*(m+4*p+3)*x^2)*(a+b*x^2+c*x^4)^(p+1),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && GtQ[m,3] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

2: $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,p<-1$

Derivation: Trinomial recurrence 2b with $A=1$ and $B=0$

Rule 1.2.2.2.6.1.4.2: If $b^2-4\,a\,c\neq 0\,\wedge\,p<-1$, then

$$\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx\;\longrightarrow$$

$$-\frac{(d\,x)^{m+1}\left(b^2-2\,a\,c+b\,c\,x^2\right)\left(a+b\,x^2+c\,x^4\right)^{p+1}}{2\,a\,d\,(p+1)\left(b^2-4\,a\,c\right)}+$$

$$\frac{1}{2\,a\,(p+1)\left(b^2-4\,a\,c\right)}\int (d\,x)^m\left(a+b\,x^2+c\,x^4\right)^{p+1}\left(b^2\,(m+2\,p+3)-2\,a\,c\,(m+4\,p+5)+b\,c\,(m+4\,p+7)\,x^2\right)dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  -(d*x)^(m+1)*(b^2-2*a*c+b*c*x^2)*(a+b*x^2+c*x^4)^(p+1)/(2*a*d*(p+1)*(b^2-4*a*c)) +
  1/(2*a*(p+1)*(b^2-4*a*c))*
    Int[(d*x)^m*(a+b*x^2+c*x^4)^(p+1)*Simp[b^2*(m+2*p+3)-2*a*c*(m+4*p+5)+b*c*(m+4*p+7)*x^2,x],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

8: $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,m>3\,\wedge\,m+4\,p+1\neq 0$

Reference: G&R 2.160.3

Derivation: Trinomial recurrence 3a with $A=0$, $B=1$ and $m=m-n$

Note: G&R 2.174.1 is a special case of G&R 2.160.3.

Rule 1.2.2.2.6.1.5: If $b^2-4\,a\,c\neq 0\,\wedge\,m>3\,\wedge\,m+4\,p+1\neq 0$, then

$$\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx\,\longrightarrow$$

$$\frac{d^3\,(d\,x)^{m-3}\left(a+b\,x^2+c\,x^4\right)^{p+1}}{c\,(m+4\,p+1)}-\frac{d^4}{c\,(m+4\,p+1)}\int (d\,x)^{m-4}\left(a\,(m-3)+b\,(m+2\,p-1)\,x^2\right)\left(a+b\,x^2+c\,x^4\right)^p dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  d^3*(d*x)^(m-3)*(a+b*x^2+c*x^4)^(p+1)/(c*(m+4*p+1)) -
  d^4/(c*(m+4*p+1))*
    Int[(d*x)^(m-4)*Simp[a*(m-3)+b*(m+2*p-1)*x^2,x]*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,d,p},x] && NeQ[b^2-4*a*c,0] && GtQ[m,3] && NeQ[m+4*p+1,0] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

9: $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p \,dx$ when $b^2-4\,a\,c\neq 0\,\land\,m<-1$

Reference: G&R 2.160.1

Derivation: Trinomial recurrence 3b with $A=1$ and $B=0$

Note: G&R 2.161.6 is a special case of G&R 2.160.1.

Rule 1.2.2.2.6.1.6: If $b^2-4\,a\,c\neq 0\,\land\,m<-1$, then

$$\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p \,dx \longrightarrow$$

$$\frac{(d\,x)^{m+1}\left(a+b\,x^2+c\,x^4\right)^{p+1}}{a\,d\,(m+1)}-\frac{1}{a\,d^2\,(m+1)}\int (d\,x)^{m+2}\left(b\,(m+2\,p+3)+c\,(m+4\,p+5)\,x^2\right)\left(a+b\,x^2+c\,x^4\right)^p\,dx$$

Program code:

```
Int[(d_.*x_)^m_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*x^2+c*x^4)^(p+1)/(a*d*(m+1)) -
  1/(a*d^2*(m+1))*Int[(d*x)^(m+2)*(b*(m+2*p+3)+c*(m+4*p+5)*x^2)*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,d,p},x] && NeQ[b^2-4*a*c,0] && LtQ[m,-1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

10. $\int \frac{(d\,x)^m}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0$

1: $\int \frac{(d\,x)^m}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0 \land m<-1$

Reference: G&R 2.176, CRC 123

Derivation: Algebraic expansion

Basis: $\frac{(d\,z)^m}{a+b\,z+c\,z^2} == \frac{(d\,z)^m}{a} - \frac{1}{a\,d}\,\frac{(d\,z)^{m+1}\,(b+c\,z)}{a+b\,z+c\,z^2}$

Rule 1.2.2.2.6.1.7.1: If $b^2-4\,a\,c\neq 0 \land m<-1$, then

$$\int \frac{(d\,x)^m}{a+b\,x^2+c\,x^4}\,dx \longrightarrow \frac{(d\,x)^{m+1}}{a\,d\,(m+1)} - \frac{1}{a\,d^2}\int \frac{(d\,x)^{m+2}\,\left(b+c\,x^2\right)}{a+b\,x^2+c\,x^4}\,dx$$

Program code:

```
Int[(d_.*x_)^m_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  (d*x)^(m+1)/(a*d*(m+1)) -
  1/(a*d^2)*Int[(d*x)^(m+2)*(b+c*x^2)/(a+b*x^2+c*x^4),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && LtQ[m,-1]
```

2. $\int \frac{(d\,x)^m}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0 \land m>3$

1: $\int \frac{x^m}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0 \land m>5 \land m\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule 1.2.2.2.6.1.7.2.1: If $b^2-4\,a\,c\neq 0 \land m>5 \land m\in\mathbb{Z}$, then

$$\int \frac{x^m}{a+b\,x^2+c\,x^4}\,dx \longrightarrow \int \text{PolynomialDivide}\left[x^m,\,a+b\,x^2+c\,x^4,\,x\right]dx$$

Program code:

```
Int[x_^m_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  Int[PolynomialDivide[x^m,(a+b*x^2+c*x^4),x],x] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && IGtQ[m,5]
```

2: $\int \frac{(d\,x)^m}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,m>3$ **Not necessary?**

- **Reference: G&R 2.174.1, CRC 119**
- **Derivation: Algebraic expansion**
- **Basis:** $\frac{(d\,z)^m}{a+b\,z+c\,z^2}=\frac{d^2\,(d\,z)^{m-2}}{c}-\frac{d^2}{c}\,\frac{(d\,z)^{m-2}\,(a+b\,z)}{a+b\,z+c\,z^2}$
- **Rule 1.2.2.2.6.1.7.2.2: If** $b^2-4\,a\,c\neq 0\,\wedge\,m>3$**, then**

$$\int \frac{(d\,x)^m}{a+b\,x^2+c\,x^4}\,dx\,\longrightarrow\,\frac{d^3\,(d\,x)^{m-3}}{c\,(m-3)}-\frac{d^4}{c}\int \frac{(d\,x)^{m-4}\,\left(a+b\,x^2\right)}{a+b\,x^2+c\,x^4}\,dx$$

- **Program code:**

```
Int[(d_.*x_)^m_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  d^3*(d*x)^(m-3)/(c*(m-3)) - d^4/c*Int[(d*x)^(m-4)*(a+b*x^2)/(a+b*x^2+c*x^4),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && GtQ[m,3]
```

3. $\int \frac{x^m}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,m\in\mathbb{Z}^+\,\wedge\,1\leq m<4\,\wedge\,b^2-4\,a\,c\not>0$

1: $\int \frac{x^2}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c<0\,\wedge\,a\,c>0$

- **Derivation: Algebraic expansion**
- **Basis: Let** $q\to\sqrt{\frac{a}{c}}$**, then** $\frac{x^2}{a+b\,x^2+c\,x^4}=\frac{q+x^2}{2\,(a+b\,x^2+c\,x^4)}-\frac{q-x^2}{2\,(a+b\,x^2+c\,x^4)}$
- **Note: Resulting integrands are of the form** $\frac{d+e\,x^2}{a+b\,x^2+c\,x^4}$ **where** $c\,d^2-a\,e^2=0\,\wedge\,b^2-4\,a\,c\not>0$**, for which there is rule.**
- **Rule 1.2.2.2.6.1.7.3.1: If** $b^2-4\,a\,c<0\,\wedge\,a\,c>0$**, let** $q\to\sqrt{\frac{a}{c}}$**, then**

$$\int \frac{x^2}{a+b\,x^2+c\,x^4}\,dx\,\longrightarrow\,\frac{1}{2}\int \frac{q+x^2}{a+b\,x^2+c\,x^4}\,dx-\frac{1}{2}\int \frac{q-x^2}{a+b\,x^2+c\,x^4}\,dx$$

- **Program code:**

```
Int[x_^2/(a_+b_.*x_^2+c_.*x_^4), x_Symbol] :=
  With[{q=Rt[a/c,2]},
  1/2*Int[(q+x^2)/(a+b*x^2+c*x^4),x] - 1/2*Int[(q-x^2)/(a+b*x^2+c*x^4),x]] /;
FreeQ[{a,b,c},x] && LtQ[b^2-4*a*c,0] && PosQ[a*c]
```

2: $\int \frac{x^m}{a+b x^2+c x^4} \, dx$ when $b^2-4ac \neq 0 \land m \in \mathbb{Z}^+ \land 3 \leq m < 4 \land b^2-4ac \ngtr 0$

Derivation: Algebraic expansion

- **Basis:** If $q \to \sqrt{\frac{a}{c}}$ and $r \to \sqrt{2q-\frac{b}{c}}$, then $\frac{z^3}{a+b z^2+c z^4} = \frac{q+r z}{2 c r (q+r z+z^2)} - \frac{q-r z}{2 c r (q-r z+z^2)}$
- **Note:** If $(a \mid b \mid c) \in \mathbb{R} \land b^2-4ac<0$, then $\frac{a}{c}>0$ and $2\sqrt{\frac{a}{c}}-\frac{b}{c}>0$.
- **Rule 1.2.2.2.6.1.7.3.2:** If $b^2-4ac \neq 0 \land m \in \mathbb{Z}^+ \land 3 \leq m < 4 \land b^2-4ac \ngtr 0$, let $q \to \sqrt{\frac{a}{c}}$ and $r \to \sqrt{2q-\frac{b}{c}}$, then

$$\int \frac{x^m}{a+b x^2+c x^4} \, dx \longrightarrow \frac{1}{2 c r} \int \frac{x^{m-3} (q+r x)}{q+r x+x^2} \, dx - \frac{1}{2 c r} \int \frac{x^{m-3} (q-r x)}{q-r x+x^2} \, dx$$

Program code:

```
Int[x_^m_./(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[a/c,2]},
  With[{r=Rt[2*q-b/c,2]},
  1/(2*c*r)*Int[x^(m-3)*(q+r*x)/(q+r*x+x^2),x] -
  1/(2*c*r)*Int[x^(m-3)*(q-r*x)/(q-r*x+x^2),x]]] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && GeQ[m,3] && LtQ[m,4] && NegQ[b^2-4*a*c]
```

3: $\int \frac{x^m}{a + b x^2 + c x^4} dx$ when $b^2 - 4 a c \neq 0 \wedge m \in \mathbb{Z}^+ \wedge 1 \leq m < 3 \wedge b^2 - 4 a c \not> 0$

Derivation: Algebraic expansion

- **Basis: If** $q \to \sqrt{\frac{a}{c}}$ **and** $r \to \sqrt{2 q - \frac{b}{c}}$**, then** $\frac{z}{a+b z^2+c z^4} == \frac{1}{2 c r (q-r z+z^2)} - \frac{1}{2 c r (q+r z+z^2)}$
- **Note: If** $(a \mid b \mid c) \in \mathbb{R} \wedge b^2 - 4 a c < 0$**, then** $\frac{a}{c} > 0$ **and** $2 \sqrt{\frac{a}{c}} - \frac{b}{c} > 0$.
- **Rule 1.2.2.2.6.1.7.3.3: If** $b^2 - 4 a c \neq 0 \wedge m \in \mathbb{Z}^+ \wedge 1 \leq m < 3 \wedge b^2 - 4 a c \not> 0$**, let** $q \to \sqrt{\frac{a}{c}}$ **and** $r \to \sqrt{2 q - \frac{b}{c}}$**, then**

$$\int \frac{x^m}{a + b x^2 + c x^4} dx \longrightarrow \frac{1}{2 c r} \int \frac{x^{m-1}}{q - r x + x^2} dx - \frac{1}{2 c r} \int \frac{x^{m-1}}{q + r x + x^2} dx$$

Program code:

```
Int[x_^m_./(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[a/c,2]},
  With[{r=Rt[2*q-b/c,2]},
  1/(2*c*r)*Int[x^(m-1)/(q-r*x+x^2),x] - 1/(2*c*r)*Int[x^(m-1)/(q+r*x+x^2),x]]] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && GeQ[m,1] && LtQ[m,3] && NegQ[b^2-4*a*c]
```

4: $\int \frac{(d\,x)^m}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\;\wedge\;m\geq 2$

Reference: G&R 2.161.1a & G&R 2.161.3

Derivation: Algebraic expansion

- **Basis: Let** $q\to\sqrt{b^2-4\,a\,c}$, **then** $\frac{(d\,z)^m}{a+b\,z+c\,z^2}==\frac{d}{2}\left(\frac{b}{q}+1\right)\frac{(d\,z)^{m-1}}{\frac{b}{2}+\frac{q}{2}+c\,z}-\frac{d}{2}\left(\frac{b}{q}-1\right)\frac{(d\,z)^{m-1}}{\frac{b}{2}-\frac{q}{2}+c\,z}$

- **Rule 1.2.2.2.6.1.7.4: If** $b^2-4\,a\,c\neq 0\;\wedge\;m\geq 2$, **let** $q\to\sqrt{b^2-4\,a\,c}$, **then**

$$\int \frac{(d\,x)^m}{a+b\,x^2+c\,x^4}\,dx\;\longrightarrow\;\frac{d^2}{2}\left(\frac{b}{q}+1\right)\int\frac{(d\,x)^{m-2}}{\frac{b}{2}+\frac{q}{2}+c\,x^2}\,dx-\frac{d^2}{2}\left(\frac{b}{q}-1\right)\int\frac{(d\,x)^{m-2}}{\frac{b}{2}-\frac{q}{2}+c\,x^2}\,dx$$

Program code:

```
Int[(d_.*x_)^m_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  d^2/2*(b/q+1)*Int[(d*x)^(m-2)/(b/2+q/2+c*x^2),x] -
  d^2/2*(b/q-1)*Int[(d*x)^(m-2)/(b/2-q/2+c*x^2),x]] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && GeQ[m,2]
```

5: $\int \frac{(d\,x)^m}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0$

Reference: G&R 2.161.1a

Derivation: Algebraic expansion

- **Basis: Let** $q\to\sqrt{b^2-4\,a\,c}$, **then** $\frac{1}{a+b\,z+c\,z^2}==\frac{c}{q}\,\frac{1}{\frac{b}{2}-\frac{q}{2}+c\,z}-\frac{c}{q}\,\frac{1}{\frac{b}{2}+\frac{q}{2}+c\,z}$

- **Rule 1.2.2.2.6.1.7.5: If** $b^2-4\,a\,c\neq 0$, **let** $q\to\sqrt{b^2-4\,a\,c}$, **then**

$$\int \frac{(d\,x)^m}{a+b\,x^2+c\,x^4}\,dx\;\longrightarrow\;\frac{c}{q}\int \frac{(d\,x)^m}{\frac{b}{2}-\frac{q}{2}+c\,x^2}\,dx-\frac{c}{q}\int \frac{(d\,x)^m}{\frac{b}{2}+\frac{q}{2}+c\,x^2}\,dx$$

Program code:

```
Int[(d_.*x_)^m_./(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  c/q*Int[(d*x)^m/(b/2-q/2+c*x^2),x] - c/q*Int[(d*x)^m/(b/2+q/2+c*x^2),x]] /;
FreeQ[{a,b,c,d,m},x] && NeQ[b^2-4*a*c,0]
```

11. $\int \frac{x^2}{\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4 a c \neq 0$

1. $\int \frac{x^2}{\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4 a c > 0$

1: $\int \frac{x^2}{\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4 a c > 0 \wedge c < 0$

Derivation: Algebraic expansion

- **Basis: If** $b^2-4 a c > 0 \wedge c < 0$, **let** $q \to \sqrt{b^2-4 a c}$, **then** $\sqrt{a+b x^2+c x^4} == \frac{1}{2\sqrt{-c}}\sqrt{b+q+2 c x^2}\,\sqrt{-b+q-2 c x^2}$
- **Rule 1.2.2.2.6.1.8.1.1: If** $b^2-4 a c > 0 \wedge c < 0$, **let** $q \to \sqrt{b^2-4 a c}$, **then**

$$\int \frac{x^2}{\sqrt{a+b x^2+c x^4}}\,dx \longrightarrow 2\sqrt{-c}\int \frac{x^2}{\sqrt{b+q+2 c x^2}\,\sqrt{-b+q-2 c x^2}}\,dx$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  2*Sqrt[-c]*Int[x^2/(Sqrt[b+q+2*c*x^2]*Sqrt[-b+q-2*c*x^2]),x]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0] && LtQ[c,0]
```

2. $\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land c \not< 0$

1: $\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \bigwedge \frac{c}{a}>0 \bigwedge \frac{b}{a}<0$

Derivation: Algebraic expansion

- **Rule 1.2.2.2.6.1.8.1.2.1: If** $b^2-4 a c>0 \bigwedge \frac{c}{a}>0 \bigwedge \frac{b}{a}<0$**, let** $q \to \sqrt{\frac{c}{a}}$**, then**

$$\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx \longrightarrow \frac{1}{q} \int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx - \frac{1}{q} \int \frac{1-q x^2}{\sqrt{a+b x^2+c x^4}} \, dx$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[c/a,2]},
  1/q*Int[1/Sqrt[a+b*x^2+c*x^4],x] - 1/q*Int[(1-q*x^2)/Sqrt[a+b*x^2+c*x^4],x]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0] && GtQ[c/a,0] && LtQ[b/a,0]
```

2: $\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land a<0 \land c>0$

Derivation: Algebraic expansion

- **Rule 1.2.2.2.6.1.8.1.2.2: If** $b^2-4 a c>0 \land a<0 \land c>0$**, let** $q \to \sqrt{b^2-4 a c}$**, then**

$$\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx \longrightarrow -\frac{b-q}{2 c} \int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx + \frac{1}{2 c} \int \frac{b-q+2 c x^2}{\sqrt{a+b x^2+c x^4}} \, dx$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  -(b-q)/(2*c)*Int[1/Sqrt[a+b*x^2+c*x^4],x] + 1/(2*c)*Int[(b-q+2*c*x^2)/Sqrt[a+b*x^2+c*x^4],x]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0] && LtQ[a,0] && GtQ[c,0]
```

3. $\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \bigwedge \frac{b \pm \sqrt{b^2-4 a c}}{a}>0$

1: $\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \bigwedge \frac{b+\sqrt{b^2-4 a c}}{a}>0$

Reference: G&R 3.153.1+

- **Rule 1.2.2.2.6.1.8.1.2.3.1: If** $b^2-4 a c>0$**, let** $q \to \sqrt{b^2-4 a c}$ **, if** $\frac{b+q}{a}>0$**, then**

$$\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx \longrightarrow \frac{x \left(b+q+2 c x^2\right)}{2 c \sqrt{a+b x^2+c x^4}}-\frac{\sqrt{\frac{b+q}{2 a}} \left(2 a+(b+q) x^2\right) \sqrt{\frac{2 a+(b-q) x^2}{2 a+(b+q) x^2}}}{2 c \sqrt{a+b x^2+c x^4}} \text{EllipticE}\left[\text{ArcTan}\left[\sqrt{\frac{b+q}{2 a}} x\right], \frac{2 q}{b+q}\right]$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  x*(b+q+2*c*x^2)/(2*c*Sqrt[a+b*x^2+c*x^4]) -
  Rt[(b+q)/(2*a),2]*(2*a+(b+q)*x^2)*Sqrt[(2*a+(b-q)*x^2)/(2*a+(b+q)*x^2)]/(2*c*Sqrt[a+b*x^2+c*x^4])*
    EllipticE[ArcTan[Rt[(b+q)/(2*a),2]*x],2*q/(b+q)] /;
 PosQ[(b+q)/a] && Not[PosQ[(b-q)/a] && SimplerSqrtQ[(b-q)/(2*a),(b+q)/(2*a)]]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

2: $\int \frac{x^2}{\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4 a c>0 \bigwedge \frac{b-\sqrt{b^2-4 a c}}{a}>0$

Reference: G&R 3.153.1-

- **Rule 1.2.2.2.6.1.8.1.2.3.2: If** $b^2-4 a c>0$**, let** $q\to\sqrt{b^2-4 a c}$**, if** $\frac{b-q}{a}>0$ **then**

$$\int \frac{x^2}{\sqrt{a+b x^2+c x^4}}\,dx \longrightarrow \frac{x\left(b-q+2 c x^2\right)}{2 c \sqrt{a+b x^2+c x^4}}-\frac{\sqrt{\frac{b-q}{2 a}}\,\left(2 a+(b-q) x^2\right) \sqrt{\frac{2 a+(b+q) x^2}{2 a+(b-q) x^2}}}{2 c \sqrt{a+b x^2+c x^4}}\,\text{EllipticE}\left[\text{ArcTan}\left[\sqrt{\frac{b-q}{2 a}}\,x\right],-\frac{2 q}{b-q}\right]$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  x*(b-q+2*c*x^2)/(2*c*Sqrt[a+b*x^2+c*x^4]) -
  Rt[(b-q)/(2*a),2]*(2*a+(b-q)*x^2)*Sqrt[(2*a+(b+q)*x^2)/(2*a+(b-q)*x^2)]/(2*c*Sqrt[a+b*x^2+c*x^4])*
    EllipticE[ArcTan[Rt[(b-q)/(2*a),2]*x],-2*q/(b-q)] /;
 PosQ[(b-q)/a]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

4. $\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \wedge \frac{b \pm \sqrt{b^2-4 a c}}{a} \ngtr 0$

1: $\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \wedge \frac{b+\sqrt{b^2-4 a c}}{a} \ngtr 0$

Derivation: Algebraic expansion

- **Rule 1.4.1.8.1.2.4.1: If $b^2-4 a c>0$, let $q \to \sqrt{b^2-4 a c}$, if $\frac{b+q}{a} \ngtr 0$ then**

$$\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx \longrightarrow -\frac{b+q}{2 c} \int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx+\frac{1}{2 c} \int \frac{b+q+2 c x^2}{\sqrt{a+b x^2+c x^4}} \, dx$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  -(b+q)/(2*c)*Int[1/Sqrt[a+b*x^2+c*x^4],x] + 1/(2*c)*Int[(b+q+2*c*x^2)/Sqrt[a+b*x^2+c*x^4],x] /;
 NegQ[(b+q)/a] && Not[NegQ[(b-q)/a] && SimplerSqrtQ[-(b-q)/(2*a),-(b+q)/(2*a)]]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

2: $\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \wedge \frac{b-\sqrt{b^2-4 a c}}{a} \ngtr 0$

Derivation: Algebraic expansion

- **Rule 1.4.1.8.1.2.4.2: If $b^2-4 a c>0$, let $q \to \sqrt{b^2-4 a c}$, if $\frac{b-q}{a} \ngtr 0$ then**

$$\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx \longrightarrow -\frac{b-q}{2 c} \int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx+\frac{1}{2 c} \int \frac{b-q+2 c x^2}{\sqrt{a+b x^2+c x^4}} \, dx$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  -(b-q)/(2*c)*Int[1/Sqrt[a+b*x^2+c*x^4],x] + 1/(2*c)*Int[(b-q+2*c*x^2)/Sqrt[a+b*x^2+c*x^4],x] /;
 NegQ[(b-q)/a]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

2. $\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c \neq 0$

1: $\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c \neq 0 \bigwedge \frac{c}{a}>0$

Derivation: Algebraic expansion

- **Rule 1.2.2.2.6.1.8.2.1: If** $b^2-4 a c \neq 0 \bigwedge \frac{c}{a}>0$**, let** $q \to \sqrt{\frac{c}{a}}$**, then**

$$\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx \longrightarrow \frac{1}{q} \int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx - \frac{1}{q} \int \frac{1-q x^2}{\sqrt{a+b x^2+c x^4}} \, dx$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[c/a,2]},
  1/q*Int[1/Sqrt[a+b*x^2+c*x^4],x] - 1/q*Int[(1-q*x^2)/Sqrt[a+b*x^2+c*x^4],x]] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && PosQ[c/a]
```

2: $\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c \neq 0 \bigwedge \frac{c}{a} \not> 0$

Derivation: Piecewise constant extraction

- **Basis: If** $q \to \sqrt{b^2-4 a c}$ **, then** $\partial_x \frac{\sqrt{1+\frac{2 c x^2}{b-q}} \sqrt{1+\frac{2 c x^2}{b+q}}}{\sqrt{a+b x^2+c x^4}} == 0$

- **Rule 1.2.2.2.6.1.8.2.2: If** $b^2-4 a c \neq 0 \bigwedge \frac{c}{a} \not> 0$**, let** $q \to \sqrt{b^2-4 a c}$ **, then**

$$\int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx \longrightarrow \frac{\sqrt{1+\frac{2 c x^2}{b-q}} \sqrt{1+\frac{2 c x^2}{b+q}}}{\sqrt{a+b x^2+c x^4}} \int \frac{x^2}{\sqrt{1+\frac{2 c x^2}{b-q}} \sqrt{1+\frac{2 c x^2}{b+q}}} \, dx$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]/Sqrt[a+b*x^2+c*x^4]*
    Int[x^2/(Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]),x]] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && NegQ[c/a]
```

12: $\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \frac{\left(a+b\,x^2+c\,x^4\right)^p}{\left(1+\frac{2\,c\,x^2}{b+\sqrt{b^2-4\,a\,c}}\right)^p \left(1+\frac{2\,c\,x^2}{b-\sqrt{b^2-4\,a\,c}}\right)^p} == 0$
- **Rule 1.2.2.2.10:**

$$\int (d\,x)^m \left(a+b\,x^2+c\,x^4\right)^p dx \longrightarrow \frac{a^{\text{IntPart}[p]} \left(a+b\,x^2+c\,x^4\right)^{\text{FracPart}[p]}}{\left(1+\frac{2\,c\,x^2}{b+\sqrt{b^2-4\,a\,c}}\right)^{\text{FracPart}[p]} \left(1+\frac{2\,c\,x^2}{b-\sqrt{b^2-4\,a\,c}}\right)^{\text{FracPart}[p]}} \int (d\,x)^m \left(1+\frac{2\,c\,x^2}{b+\sqrt{b^2-4\,a\,c}}\right)^p \left(1+\frac{2\,c\,x^2}{b-\sqrt{b^2-4\,a\,c}}\right)^p dx$$

- **Program code:**

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  a^IntPart[p]*(a+b*x^2+c*x^4)^FracPart[p]/
    ((1+2*c*x^2/(b+Rt[b^2-4*a*c,2]))^FracPart[p]*(1+2*c*x^2/(b-Rt[b^2-4*a*c,2]))^FracPart[p])*
    Int[(d*x)^m*(1+2*c*x^2/(b+Sqrt[b^2-4*a*c]))^p*(1+2*c*x^2/(b-Sqrt[b^2-4*a*c]))^p,x] /;
FreeQ[{a,b,c,d,m,p},x]
```

S: $\int u^m \left(a+b\,v^2+c\,v^4\right)^p dx$ when $v == d+e\,x \land u == f\,v$

- **Derivation: Integration by substitution and piecewise constant extraction**
- **Basis: If** $u == f\,v$**, then** $\partial_x \frac{u^m}{v^m} == 0$
- **Rule 1.2.2.2.S: If** $v == d+e\,x \land u == f\,v$**, then**

$$\int u^m \left(a+b\,v^2+c\,v^4\right)^p dx \longrightarrow \frac{u^m}{e\,v^m} \text{Subst}\left[\int x^m \left(a+b\,x^2+c\,x^4\right)^p dx,\ x,\ v\right]$$

- **Program code:**

```
Int[u_^m_.*(a_.+b_.*v_^2+c_.*v_^4)^p_.,x_Symbol] :=
  u^m/(Coefficient[v,x,1]*v^m)*Subst[Int[x^m*(a+b*x^2+c*x^(2*2))^p,x],x,v] /;
FreeQ[{a,b,c,m,p},x] && LinearPairQ[u,v,x]
```