Rules for integrands of the form $(a+b x)^m (c+d x)^n (e+f x)^p$

when $b c-a d\neq 0 \bigwedge b e-a f\neq 0 \bigwedge d e-c f\neq 0$

0: $\int \frac{(e+f x)^p}{(a+b x) (c+d x)} \, dx$ when $p\in\mathbb{Z}$

- **Derivation: Algebraic expansion**
- **Rule 1.1.1.3.0: If $p\in\mathbb{Z}$, then**

$$\int \frac{(e+f x)^p}{(a+b x) (c+d x)} \, dx \longrightarrow \int \text{ExpandIntegrand}\left[\frac{(e+f x)^p}{(a+b x) (c+d x)}, x\right] dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^p_./((a_.+b_.*x_)*(c_.+d_.*x_)),x_Symbol] :=
  Int[ExpandIntegrand[(e+f*x)^p/((a+b*x)*(c+d*x)),x],x] /;
FreeQ[{a,b,c,d,e,f},x] && IntegerQ[p]
```

1: $\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx$ when $b c+a d=0 \bigwedge n=m \bigwedge m\in\mathbb{Z}$

- **Derivation: Algebraic simplification**
- **Basis: If $b c+a d=0 \wedge m\in\mathbb{Z}$, then $(a+b x)^m (c+d x)^m = \left(a c+b d x^2\right)^m$**
- **Rule 1.1.1.3.1: If $b c+a d=0 \wedge n=m \wedge m\in\mathbb{Z}$, then**

$$\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx \longrightarrow \int \left(a c+b d x^2\right)^m (e+f x)^p \, dx$$

- **Program code:**

```
Int[(a_+b_.*x_)^m_.*(c_+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  Int[(a*c+b*d*x^2)^m*(e+f*x)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && EqQ[b*c+a*d,0] && EqQ[n,m] && IntegerQ[m]
```

2. $\int (a+b x) (c+d x)^n (e+f x)^p \, dx$

1: $\int (a+b x) (c+d x)^n (e+f x)^p \, dx$ when $n+p+2\neq 0 \bigwedge a d f (n+p+2)-b (d e (n+1)+c f (p+1))=0$

- **Derivation: Quadratic recurrence 2b with $c=0$: linear recurrence 2 with $a d f (n+p+2)-b (d e (n+1)+c f (p+1))=0$**
- **Rule 1.1.1.3.2.1: If $n+p+2\neq 0 \bigwedge a d f (n+p+2)-b (d e (n+1)+c f (p+1))=0$, then**

$$\int (a+b x) (c+d x)^n (e+f x)^p \, dx \longrightarrow \frac{b (c+d x)^{n+1} (e+f x)^{p+1}}{d f (n+p+2)}$$

Program code:

```
Int[(a_.+b_.*x_)*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  b*(c+d*x)^(n+1)*(e+f*x)^(p+1)/(d*f*(n+p+2)) /;
FreeQ[{a,b,c,d,e,f,n,p},x] && NeQ[n+p+2,0] && EqQ[a*d*f*(n+p+2)-b*(d*e*(n+1)+c*f*(p+1)),0]
```

2: $\int (a+b x) (c+d x)^n (e+f x)^p \, dx$ when $b c-a d \neq 0 \land ((n \mid p) \in \mathbb{Z}^- \lor p == 1 \lor p \in \mathbb{Z}^+ \land (n \notin \mathbb{Z} \lor 9 p+5 (n+2) \leq 0 \lor n+p+1 \geq 0))$

Derivation: Algebraic expansion

Rule 1.1.1.3.2.2: If $b c-a d \neq 0 \land ((n \mid p) \in \mathbb{Z}^- \lor p == 1 \lor p \in \mathbb{Z}^+ \land (n \notin \mathbb{Z} \lor 9 p+5 (n+2) \leq 0 \lor n+p+1 \geq 0))$**, then**

$$\int (a+b x) (c+d x)^n (e+f x)^p \, dx \longrightarrow \int \text{ExpandIntegrand}[(a+b x) (c+d x)^n (e+f x)^p, x] \, dx$$

Program code:

```
Int[(a_+b_.*x_)*(d_.*x_)^n_.*(e_+f_.*x_)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x)*(d*x)^n*(e+f*x)^p,x],x] /;
FreeQ[{a,b,d,e,f,n},x] && IGtQ[p,0] && EqQ[b*e+a*f,0] && Not[ILtQ[n+p+2,0] && GtQ[n+2*p,0]]
```

```
Int[(a_+b_.*x_)*(d_.*x_)^n_.*(e_+f_.*x_)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x)*(d*x)^n*(e+f*x)^p,x],x] /;
FreeQ[{a,b,d,e,f,n},x] && IGtQ[p,0] && (NeQ[n,-1] || EqQ[p,1]) && NeQ[b*e+a*f,0] &&
  (Not[IntegerQ[n]] || LtQ[9*p+5*n,0] || GeQ[n+p+1,0] || GeQ[n+p+2,0] && RationalQ[a,b,d,e,f]) && (NeQ[n+p+3,0] || EqQ[p,1])
```

```
Int[(a_.+b_.*x_)*(c_+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x)*(c+d*x)^n*(e+f*x)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && NeQ[b*c-a*d,0] &&
  (ILtQ[n,0] && ILtQ[p,0] || EqQ[p,1] ||
    IGtQ[p,0] && (Not[IntegerQ[n]] || LeQ[9*p+5*(n+2),0] || GeQ[n+p+1,0] || GeQ[n+p+2,0] && RationalQ[a,b,c,d,e,f]))
```

3: $\int (a+b\,x)\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $p<-1\ \land\ (n\nless -1\ \lor\ p\in\mathbb{Z})$

Derivation: Quadratic recurrence 2b with $c=0$

Derivation: Quadratic recurrence 3b with $c=0$, $n=p$ and $p=n$

- **Note: If n and p are both negative and one is an integer, best to drive that integer exponent toward -1 since the terms of the antiderivative of $\frac{(a+b\,x)^m}{c+d\,x}$ are of the form $g\,(a+b\,x)^k$.**

Rule 1.1.1.3.2.3: If $p<-1\ \land\ (n\nless -1\ \lor\ p\in\mathbb{Z})$, then

$$\int (a+b\,x)\,(c+d\,x)^n\,(e+f\,x)^p\,dx\longrightarrow$$
$$-\frac{(b\,e-a\,f)\,(c+d\,x)^{n+1}\,(e+f\,x)^{p+1}}{f\,(p+1)\,(c\,f-d\,e)}-\frac{a\,d\,f\,(n+p+2)-b\,(d\,e\,(n+1)+c\,f\,(p+1))}{f\,(p+1)\,(c\,f-d\,e)}\int (c+d\,x)^n\,(e+f\,x)^{p+1}\,dx$$

Program code:

```
Int[(a_.+b_.*x_)*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  -(b*e-a*f)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/(f*(p+1)*(c*f-d*e)) -
  (a*d*f*(n+p+2)-b*(d*e*(n+1)+c*f*(p+1)))/(f*(p+1)*(c*f-d*e))*Int[(c+d*x)^n*(e+f*x)^(p+1),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && LtQ[p,-1] &&
  (Not[LtQ[n,-1]] || IntegerQ[p] || Not[IntegerQ[n]] || Not[EqQ[e,0]] || Not[EqQ[c,0]] || LtQ[p,n]]]])
```

```
Int[(a_.+b_.*x_)*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  -(b*e-a*f)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/(f*(p+1)*(c*f-d*e)) -
  (a*d*f*(n+p+2)-b*(d*e*(n+1)+c*f*(p+1)))/(f*(p+1)*(c*f-d*e))*Int[(c+d*x)^n*(e+f*x)^Simplify[p+1],x] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && Not[RationalQ[p]] && SumSimplerQ[p,1]
```

4: $\int (a+b x) (c+d x)^n (e+f x)^p \, dx$ when $n+p+2 \neq 0$

Derivation: Quadratic recurrence 2b with $c = 0$: linear recurrence 2

Rule 1.1.1.3.2.4: If $n+p+2 \neq 0$, then

$$\int (a+b x) (c+d x)^n (e+f x)^p \, dx \longrightarrow$$
$$\frac{b (c+d x)^{n+1} (e+f x)^{p+1}}{d f (n+p+2)} + \frac{a d f (n+p+2) - b (d e (n+1) + c f (p+1))}{d f (n+p+2)} \int (c+d x)^n (e+f x)^p \, dx$$

Program code:

```
Int[(a_.+b_.*x_)*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  b*(c+d*x)^(n+1)*(e+f*x)^(p+1)/(d*f*(n+p+2)) +
  (a*d*f*(n+p+2)-b*(d*e*(n+1)+c*f*(p+1)))/(d*f*(n+p+2))*Int[(c+d*x)^n*(e+f*x)^p,x] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && NeQ[n+p+2,0]
```

3: $\int (a+b x)^2 (c+d x)^n (e+f x)^p \, dx$ when

$n+p+2 \neq 0 \land n+p+3 \neq 0 \land$
$d f (n+p+2) \left(a^2 d f (n+p+3) - b (b c e + a (d e (n+1) + c f (p+1)))\right) - b (d e (n+1) + c f (p+1)) (a d f (n+p+4) - b (d e (n+2) + c f (p+2))) == 0$

Derivation: Nondegenerate trilinear recurrence 2 with $A = a$ and $B = b$: quadratic recurrence 2b with $c = 0$: linear recurrence 2 with $a d f (n+p+2) - b (d e (n+1) + c f (p+1)) == 0$

Rule 1.1.1.3.3: If $n+p+2 \neq 0 \land n+p+3 \neq 0 \land$ $d f (n+p+2) \left(a^2 d f (n+p+3) - b (b c e + a (d e (n+1) + c f (p+1)))\right) - b (d e (n+1) + c f (p+1)) (a d f (n+p+4) - b (d e (n+2) + c f (p+2))) == 0$, then

$$\int (a+b x)^2 (c+d x)^n (e+f x)^p \, dx \longrightarrow$$
$$\left(b (c+d x)^{n+1} (e+f x)^{p+1} (2 a d f (n+p+3) - b (d e (n+2) + c f (p+2)) + b d f (n+p+2) x)\right) \Big/ \left(d^2 f^2 (n+p+2) (n+p+3)\right)$$

Program code:

```
Int[(a_.+b_.*x_)^2*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  b*(c+d*x)^(n+1)*(e+f*x)^(p+1)*(2*a*d*f*(n+p+3)-b*(d*e*(n+2)+c*f*(p+2))+b*d*f*(n+p+2)*x)/(d^2*f^2*(n+p+2)*(n+p+3)) /;
FreeQ[{a,b,c,d,e,f,n,p},x] && NeQ[n+p+2,0] && NeQ[n+p+3,0] &&
  EqQ[d*f*(n+p+2)*(a^2*d*f*(n+p+3)-b*(b*c*e+a*(d*e*(n+1)+c*f*(p+1))))-b*(d*e*(n+1)+c*f*(p+1))*(a*d*f*(n+p+4)-b*(d*e*(n+2)+c*f*(p+2)
```

4: $\int (a+b\,x)^m\,(c+d\,x)^n\,(f\,x)^p\,dx$ when $b\,c+a\,d==0\,\wedge\,m-n==1$

Derivation: Algebraic expansion

Note: Integrals of this form can be expressed as the sum of two hypergeometric functions.

Rule 1.1.1.3.4: If $b\,c+a\,d==0\,\wedge\,m-n==1$, then

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(f\,x)^p\,dx\,\longrightarrow\,a\int (a+b\,x)^n\,(c+d\,x)^n\,(f\,x)^p\,dx+\frac{b}{f}\int (a+b\,x)^n\,(c+d\,x)^n\,(f\,x)^{p+1}\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(f_.*x_)^p_.,x_Symbol] :=
  a*Int[(a+b*x)^n*(c+d*x)^n*(f*x)^p,x] + b/f*Int[(a+b*x)^n*(c+d*x)^n*(f*x)^(p+1),x] /;
FreeQ[{a,b,c,d,f,m,n,p},x] && EqQ[b*c+a*d,0] && EqQ[m-n-1,0] && Not[RationalQ[p]] && Not[IGtQ[m,0]] && NeQ[m+n+p+2,0]
```

5. $\int \frac{(e+f\,x)^p}{(a+b\,x)\,(c+d\,x)}\,dx$ when $p\notin\mathbb{Z}$

1. $\int \frac{(e+f\,x)^p}{(a+b\,x)\,(c+d\,x)}\,dx$ when $p>0$

1: $\int \frac{(e+f\,x)^p}{(a+b\,x)\,(c+d\,x)}\,dx$ when $0<p<1$

Derivation: Algebraic expansion

Basis: $\frac{e+f\,x}{(a+b\,x)\,(c+d\,x)}==\frac{b\,e-a\,f}{(b\,c-a\,d)\,(a+b\,x)}-\frac{d\,e-c\,f}{(b\,c-a\,d)\,(c+d\,x)}$

Rule 1.1.1.3.5.1.1: If $0<p<1$, then

$$\int \frac{(e+f\,x)^p}{(a+b\,x)\,(c+d\,x)}\,dx\,\longrightarrow\,\frac{b\,e-a\,f}{b\,c-a\,d}\int \frac{(e+f\,x)^{p-1}}{a+b\,x}\,dx-\frac{d\,e-c\,f}{b\,c-a\,d}\int \frac{(e+f\,x)^{p-1}}{c+d\,x}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^p_./((a_.+b_.*x_)*(c_.+d_.*x_)),x_Symbol] :=
  (b*e-a*f)/(b*c-a*d)*Int[(e+f*x)^(p-1)/(a+b*x),x] -
  (d*e-c*f)/(b*c-a*d)*Int[(e+f*x)^(p-1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f},x] && LtQ[0,p,1]
```

2: $\int \frac{(e+f\,x)^p}{(a+b\,x)\,(c+d\,x)}\,dx$ when $p>1$

Derivation: Nondegenerate trilinear recurrence 2 with $A = a$ and $B = b$

Rule 1.1.1.3.5.1.2: If $p>1$, then

$$\int \frac{(e+f\,x)^p}{(a+b\,x)\,(c+d\,x)}\,dx \longrightarrow \frac{f\,(e+f\,x)^{p-1}}{b\,d\,(p-1)}+\frac{1}{b\,d}\int \frac{\left(b\,d\,e^2-a\,c\,f^2+f\,(2\,b\,d\,e-b\,c\,f-a\,d\,f)\,x\right)\,(e+f\,x)^{p-2}}{(a+b\,x)\,(c+d\,x)}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^p_/((a_.+b_.*x_)*(c_.+d_.*x_)),x_Symbol] :=
  f*(e+f*x)^(p-1)/(b*d*(p-1)) +
 1/(b*d)*Int[(b*d*e^2-a*c*f^2+f*(2*b*d*e-b*c*f-a*d*f)*x)*(e+f*x)^(p-2)/((a+b*x)*(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && GtQ[p,1]
```

2: $\int \frac{(e+f\,x)^p}{(a+b\,x)\,(c+d\,x)}\,dx$ when $p<-1$

Derivation: Nondegenerate trilinear recurrence 3 with $A = 1$ and $B = 0$

Rule 1.1.1.3.5.2: If $p<-1$, then

$$\int \frac{(e+f\,x)^p}{(a+b\,x)\,(c+d\,x)}\,dx \longrightarrow$$

$$\frac{f\,(e+f\,x)^{p+1}}{(p+1)\,(b\,e-a\,f)\,(d\,e-c\,f)}+\frac{1}{(b\,e-a\,f)\,(d\,e-c\,f)}\int \frac{(b\,d\,e-b\,c\,f-a\,d\,f-b\,d\,f\,x)\,(e+f\,x)^{p+1}}{(a+b\,x)\,(c+d\,x)}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^p_/((a_.+b_.*x_)*(c_.+d_.*x_)),x_Symbol] :=
  f*(e+f*x)^(p+1)/((p+1)*(b*e-a*f)*(d*e-c*f)) +
  1/((b*e-a*f)*(d*e-c*f))*Int[(b*d*e-b*c*f-a*d*f-b*d*f*x)*(e+f*x)^(p+1)/((a+b*x)*(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && LtQ[p,-1]
```

3: $\int \frac{(e+f x)^p}{(a+b x)(c+d x)} \, dx$ when $p \notin \mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\frac{1}{(a+b x)(c+d x)} == \frac{b}{(b c-a d)(a+b x)} - \frac{d}{(b c-a d)(c+d x)}$

Rule 1.1.1.3.5.3: If $p \notin \mathbb{Z}$, then

$$\int \frac{(e+f x)^p}{(a+b x)(c+d x)} \, dx \longrightarrow \frac{b}{b c-a d} \int \frac{(e+f x)^p}{a+b x} \, dx - \frac{d}{b c-a d} \int \frac{(e+f x)^p}{c+d x} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^p_/((a_.+b_.*x_)*(c_.+d_.*x_)),x_Symbol] :=
  b/(b*c-a*d)*Int[(e+f*x)^p/(a+b*x),x] -
  d/(b*c-a*d)*Int[(e+f*x)^p/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,p},x] && Not[IntegerQ[p]]
```

6: $\int \frac{(c+d x)^n (e+f x)^p}{a+b x} \, dx$ when $n \in \mathbb{Z}^+ \land p < -1$

Derivation: Algebraic expansion

Rule 1.1.1.3.6: If $n \in \mathbb{Z}^+ \land p < -1$, then

$$\int \frac{(c+d x)^n (e+f x)^p}{a+b x} \, dx \longrightarrow$$

$$\int (e+f x)^{\text{FractionalPart}[p]} \, \text{ExpandIntegrand}\left[\frac{(c+d x)^n (e+f x)^{\text{IntegerPart}[p]}}{a+b x}, x\right] dx$$

Program code:

```
Int[(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_/(a_.+b_.*x_),x_Symbol] :=
  Int[ExpandIntegrand[(e+f*x)^FractionalPart[p],(c+d*x)^n*(e+f*x)^IntegerPart[p]/(a+b*x),x],x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[n,0] && LtQ[p,-1] && FractionQ[p]
```

7: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $(m\mid n)\in\mathbb{Z}\,\wedge\,(p\in\mathbb{Z}\,\vee\,(m>0\,\wedge\,n\geq -1))$

Derivation: Algebraic expansion

Rule 1.1.1.3.7: If $(m\mid n)\in\mathbb{Z}\,\wedge\,(p\in\mathbb{Z}\,\vee\,(m>0\,\wedge\,n\geq -1))$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\,\longrightarrow\,\int \text{ExpandIntegrand}[(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p,\,x]\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x)^m*(c+d*x)^n*(e+f*x)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,p},x] && IntegersQ[m,n] && (IntegerQ[p] || GtQ[m,0] && GeQ[n,-1])
```

8. $\int (a+b\,x)^2\,(c+d\,x)^n\,(e+f\,x)^p\,dx$

1: $\int (a+b\,x)^2\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $n<-1$

Derivation: ?

Rule 1.1.1.3.8.1: If $n<-1$**, then**

$$\int (a+b\,x)^2\,(c+d\,x)^n\,(e+f\,x)^p\,dx\,\longrightarrow$$

$$\frac{(b\,c-a\,d)^2\,(c+d\,x)^{n+1}\,(e+f\,x)^{p+1}}{d^2\,(d\,e-c\,f)\,(n+1)}-$$

$$\frac{1}{d^2\,(d\,e-c\,f)\,(n+1)}\int (c+d\,x)^{n+1}\,(e+f\,x)^p\cdot$$

$$\left(a^2\,d^2\,f\,(n+p+2)+b^2\,c\,(d\,e\,(n+1)+c\,f\,(p+1))-2\,a\,b\,d\,(d\,e\,(n+1)+c\,f\,(p+1))-b^2\,d\,(d\,e\,-c\,f)\,(n+1)\,x\right)\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^2*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  (b*c-a*d)^2*(c+d*x)^(n+1)*(e+f*x)^(p+1)/(d^2*(d*e-c*f)*(n+1)) -
  1/(d^2*(d*e-c*f)*(n+1))*Int[(c+d*x)^(n+1)*(e+f*x)^p*
    Simp[a^2*d^2*f*(n+p+2)+b^2*c*(d*e*(n+1)+c*f*(p+1))-2*a*b*d*(d*e*(n+1)+c*f*(p+1))-b^2*d*(d*e-c*f)*(n+1)*x,x],x] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && (LtQ[n,-1] || EqQ[n+p+3,0] && NeQ[n,-1] && (SumSimplerQ[n,1] || Not[SumSimplerQ[p,1]]))
```

2: $\int (a+b x)^2 (c+d x)^n (e+f x)^p \, dx$ when $n+p+3 \neq 0$

Derivation: Nondegenerate trilinear recurrence 2 with $A = a$ and $B = b$

Rule 1.1.1.3.8.2: If $n+p+3 \neq 0$, then

$$\int (a+b x)^2 (c+d x)^n (e+f x)^p \, dx \longrightarrow$$
$$\frac{b (a+b x) (c+d x)^{n+1} (e+f x)^{p+1}}{d f (n+p+3)} +$$
$$\frac{1}{d f (n+p+3)} \int (c+d x)^n (e+f x)^p \cdot$$
$$\left(a^2 d f (n+p+3) - b (b c e + a (d e (n+1) + c f (p+1))) + b (a d f (n+p+4) - b (d e (n+2) + c f (p+2))) x\right) dx$$

Program code:

```
Int[(a_.+b_.*x_)^2*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  b*(a+b*x)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/(d*f*(n+p+3)) +
  1/(d*f*(n+p+3))*Int[(c+d*x)^n*(e+f*x)^p*
    Simp[a^2*d*f*(n+p+3)-b*(b*c*e+a*(d*e*(n+1)+c*f*(p+1)))+b*(a*d*f*(n+p+4)-b*(d*e*(n+2)+c*f*(p+2)))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && NeQ[n+p+3,0]
```

9. $\int \frac{(a+b\,x)^m\,(c+d\,x)^n}{e+f\,x}\,dx$ when $m+n+1=0\,\land\,-1<m<0$

1: $\int \frac{1}{(a+b\,x)^{1/3}\,(c+d\,x)^{2/3}\,(e+f\,x)}\,dx$

Rule 1.1.1.3.9.1: Let $q=\left(\frac{d\,e-c\,f}{b\,e-a\,f}\right)^{1/3}$ then

$$\int \frac{1}{(a+b\,x)^{1/3}\,(c+d\,x)^{2/3}\,(e+f\,x)}\,dx \longrightarrow$$
$$-\frac{\sqrt{3}\,q\,\mathrm{ArcTan}\left[\frac{1}{\sqrt{3}}+\frac{2\,q\,(a+b\,x)^{1/3}}{\sqrt{3}\,(c+d\,x)^{1/3}}\right]}{d\,e-c\,f}+\frac{q\,\mathrm{Log}[e+f\,x]}{2\,(d\,e-c\,f)}-\frac{3\,q\,\mathrm{Log}\left[q\,(a+b\,x)^{1/3}-(c+d\,x)^{1/3}\right]}{2\,(d\,e-c\,f)}$$

Program code:

```
Int[1/((a_.+b_.*x_)^(1/3)*(c_.+d_.*x_)^(2/3)*(e_.+f_.*x_)),x_Symbol] :=
  With[{q=Rt[(d*e-c*f)/(b*e-a*f),3]},
  -Sqrt[3]*q*ArcTan[1/Sqrt[3]+2*q*(a+b*x)^(1/3)/(Sqrt[3]*(c+d*x)^(1/3))]/(d*e-c*f) +
  q*Log[e+f*x]/(2*(d*e-c*f)) -
  3*q*Log[q*(a+b*x)^(1/3)-(c+d*x)^(1/3)]/(2*(d*e-c*f))] /;
FreeQ[{a,b,c,d,e,f},x]
```

2: $\int \frac{1}{\sqrt{a+b\,x}\,\sqrt{c+d\,x}\,(e+f\,x)}\,dx$ when $2\,b\,d\,e-f\,(b\,c+a\,d)=0$

Derivation: Integration by substitution

Basis: If $2\,b\,d\,e-f\,(b\,c+a\,d)=0$, then $\frac{1}{\sqrt{a+b\,x}\,\sqrt{c+d\,x}\,(e+f\,x)}=b\,f\,\mathrm{Subst}\left[\frac{1}{d\,(b\,e-a\,f)^2+b\,f^2\,x^2},\,x,\,\sqrt{a+b\,x}\,\sqrt{c+d\,x}\right]\,\partial_x\left(\sqrt{a+b\,x}\,\sqrt{c+d\,x}\right)$

Rule 1.1.1.3.9.2: If $2\,b\,d\,e-f\,(b\,c+a\,d)=0$, then

$$\int \frac{1}{\sqrt{a+b\,x}\,\sqrt{c+d\,x}\,(e+f\,x)}\,dx \longrightarrow b\,f\,\mathrm{Subst}\left[\int \frac{1}{d\,(b\,e-a\,f)^2+b\,f^2\,x^2}\,dx,\,x,\,\sqrt{a+b\,x}\,\sqrt{c+d\,x}\right]$$

Program code:

```
Int[1/(Sqrt[a_.+b_.*x_]*Sqrt[c_.+d_.*x_]*(e_.+f_.*x_)),x_Symbol] :=
  b*f*Subst[Int[1/(d*(b*e-a*f)^2+b*f^2*x^2),x],x,Sqrt[a+b*x]*Sqrt[c+d*x]] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[2*b*d*e-f*(b*c+a*d),0]
```

3: $\int \frac{(a+b\,x)^m\,(c+d\,x)^n}{e+f\,x}\,dx$ when $m+n+1=0\,\land\,-1<m<0$

Derivation: Integration by substitution

Basis: If $m+n+1=0\,\land\,-1<m<0$**, let** $q=\text{Denominator}[m]$**, then** $\frac{(a+b\,x)^m\,(c+d\,x)^n}{e+f\,x}=q\,\text{Subst}\left[\frac{x^{q\,(m+1)-1}}{b\,e-a\,f-(d\,e-c\,f)\,x^q},\,x,\,\frac{(a+b\,x)^{1/q}}{(c+d\,x)^{1/q}}\right]\partial_x\frac{(a+b\,x)^{1/q}}{(c+d\,x)^{1/q}}$

Rule 1.1.1.3.9.3: If $m+n+1=0\,\land\,-1<m<0$**, let** $q=\text{Denominator}[m]$**, then**

$$\int \frac{(a+b\,x)^m\,(c+d\,x)^n}{e+f\,x}\,dx\,\longrightarrow\,q\,\text{Subst}\left[\int\frac{x^{q\,(m+1)-1}}{b\,e-a\,f-(d\,e-c\,f)\,x^q}\,dx,\,x,\,\frac{(a+b\,x)^{1/q}}{(c+d\,x)^{1/q}}\right]$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_/(e_.+f_.*x_),x_Symbol] :=
  With[{q=Denominator[m]},
  q*Subst[Int[x^(q*(m+1)-1)/(b*e-a*f-(d*e-c*f)*x^q),x],x,(a+b*x)^(1/q)/(c+d*x)^(1/q)]] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[m+n+1,0] && RationalQ[n] && LtQ[-1,m,0] && SimplerQ[a+b*x,c+d*x]
```

10: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m+n+p+2=0\,\land\,n>0$

Derivation: Nondegenerate trilinear recurrence 1 with $A=1$**,** $B=0$ **and** $m+n+p+2=0$

Rule 1.1.1.3.10: If $m+n+p+2=0\,\land\,n>0$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\,\longrightarrow$$
$$\frac{(a+b\,x)^{m+1}\,(c+d\,x)^n\,(e+f\,x)^{p+1}}{(m+1)\,(b\,e-a\,f)}-\frac{n\,(d\,e-c\,f)}{(m+1)\,(b\,e-a\,f)}\int (a+b\,x)^{m+1}\,(c+d\,x)^{n-1}\,(e+f\,x)^p\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^(p+1)/((m+1)*(b*e-a*f)) -
  n*(d*e-c*f)/((m+1)*(b*e-a*f))*Int[(a+b*x)^(m+1)*(c+d*x)^(n-1)*(e+f*x)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && EqQ[m+n+p+2,0] && GtQ[n,0] && (SumSimplerQ[m,1] || Not[SumSimplerQ[p,1]]) && NeQ[m,-1]
```

11. $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m+n+p+3=0$

1: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m+n+p+3=0\,\land\,a\,d\,f\,(m+1)+b\,c\,f\,(n+1)+b\,d\,e\,(p+1)=0\,\land\,m\neq-1$

Derivation: Nondegenerate trilinear recurrence 3 with $A=1$ and $B=0$

Rule 1.1.1.3.11.1: If $m+n+p+3=0\,\land\,a\,d\,f\,(m+1)+b\,c\,f\,(n+1)+b\,d\,e\,(p+1)=0\,\land\,m\neq-1$, then

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\;\longrightarrow\;\frac{b\,(a+b\,x)^{m+1}\,(c+d\,x)^{n+1}\,(e+f\,x)^{p+1}}{(m+1)\,(b\,c-a\,d)\,(b\,e-a\,f)}$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  b*(a+b*x)^(m+1)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/((m+1)*(b*c-a*d)*(b*e-a*f)) /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && EqQ[Simplify[m+n+p+3],0] && EqQ[a*d*f*(m+1)+b*c*f*(n+1)+b*d*e*(p+1),0] && NeQ[m,-1]
```

2: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m+n+p+3=0\,\land\,m<-1$

Derivation: Nondegenerate trilinear recurrence 3 with $A=1$ and $B=0$

Rule 1.1.1.3.11.2: If $m+n+p+3=0\,\land\,m<-1$, then

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\;\longrightarrow$$

$$\frac{b\,(a+b\,x)^{m+1}\,(c+d\,x)^{n+1}\,(e+f\,x)^{p+1}}{(m+1)\,(b\,c-a\,d)\,(b\,e-a\,f)}+\frac{a\,d\,f\,(m+1)+b\,c\,f\,(n+1)+b\,d\,e\,(p+1)}{(m+1)\,(b\,c-a\,d)\,(b\,e-a\,f)}\int (a+b\,x)^{m+1}\,(c+d\,x)^n\,(e+f\,x)^p\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  b*(a+b*x)^(m+1)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/((m+1)*(b*c-a*d)*(b*e-a*f)) +
  (a*d*f*(m+1)+b*c*f*(n+1)+b*d*e*(p+1))/((m+1)*(b*c-a*d)*(b*e-a*f))*Int[(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && EqQ[Simplify[m+n+p+3],0] && (LtQ[m,-1] || SumSimplerQ[m,1])
```

12. $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m<-1\,\wedge\,n>0$

1: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m<-1\,\wedge\,n>0\,\wedge\,p>0$

Derivation: Nondegenerate trilinear recurrence 1 with $A=e$ and $B=f$

Rule 1.1.1.3.12.1: If $m<-1\,\wedge\,n>0\,\wedge\,p>0$, then

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\;\longrightarrow$$
$$\frac{(a+b\,x)^{m+1}\,(c+d\,x)^n\,(e+f\,x)^p}{b\,(m+1)}-\frac{1}{b\,(m+1)}\int (a+b\,x)^{m+1}\,(c+d\,x)^{n-1}\,(e+f\,x)^{p-1}\,(d\,e\,n+c\,f\,p+d\,f\,(n+p)\,x)\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p/(b*(m+1)) -
  1/(b*(m+1))*Int[(a+b*x)^(m+1)*(c+d*x)^(n-1)*(e+f*x)^(p-1)*Simp[d*e*n+c*f*p+d*f*(n+p)*x,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && LtQ[m,-1] && GtQ[n,0] && GtQ[p,0] && (IntegersQ[2*m,2*n,2*p] || IntegersQ[m,n+p] || IntegersQ[p,m+n])
```

2: $\int (a+bx)^m (c+dx)^n (e+fx)^p \, dx$ when $m < -1 \land n > 1$

Derivation: ???

Rule 1.1.1.3.12.2: If $m < -1 \land n > 1$, then

$$\int (a+bx)^m (c+dx)^n (e+fx)^p \, dx \longrightarrow$$

$$\frac{(b c - a d) (a+bx)^{m+1} (c+dx)^{n-1} (e+fx)^{p+1}}{b (b e - a f) (m+1)} +$$

$$\frac{1}{b (b e - a f) (m+1)} \int (a+bx)^{m+1} (c+dx)^{n-2} (e+fx)^p \cdot$$

$$(a d (d e (n-1) + c f (p+1)) + b c (d e (m-n+2) - c f (m+p+2)) + d (a d f (n+p) + b (d e (m+1) - c f (m+n+p+1))) x) \, dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  (b*c-a*d)*(a+b*x)^(m+1)*(c+d*x)^(n-1)*(e+f*x)^(p+1)/(b*(b*e-a*f)*(m+1)) +
  1/(b*(b*e-a*f)*(m+1))*Int[(a+b*x)^(m+1)*(c+d*x)^(n-2)*(e+f*x)^p*
    Simp[a*d*(d*e*(n-1)+c*f*(p+1))+b*c*(d*e*(m-n+2)-c*f*(m+p+2))+d*(a*d*f*(n+p)+b*(d*e*(m+1)-c*f*(m+n+p+1)))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,p},x] && LtQ[m,-1] && GtQ[n,1] && (IntegersQ[2*m,2*n,2*p] || IntegersQ[m,n+p] || IntegersQ[p,m+n])
```

3: $\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx$ when $m<-1 \land n>0$

Derivation: Nondegenerate trilinear recurrence 1 with $A = 1$ and $B = 0$

Rule 1.1.1.3.12.3: If $m<-1 \land n>0$, then

$$\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx \longrightarrow$$
$$\frac{(a+b x)^{m+1} (c+d x)^n (e+f x)^{p+1}}{(m+1) (b e-a f)} -$$
$$\frac{1}{(m+1) (b e-a f)} \int (a+b x)^{m+1} (c+d x)^{n-1} (e+f x)^p (d e n+c f (m+p+2)+d f (m+n+p+2) x) \, dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^(p+1)/((m+1)*(b*e-a*f)) -
  1/((m+1)*(b*e-a*f))*Int[(a+b*x)^(m+1)*(c+d*x)^(n-1)*(e+f*x)^p*
    Simp[d*e*n+c*f*(m+p+2)+d*f*(m+n+p+2)*x,x],x] /;
FreeQ[{a,b,c,d,e,f,p},x] && LtQ[m,-1] && GtQ[n,0] && (IntegersQ[2*m,2*n,2*p] || IntegersQ[m,n+p] || IntegersQ[p,m+n])
```

13: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m>1\,\wedge\,m+n+p+1\neq 0\,\wedge\,m\in\mathbb{Z}$

Derivation: Nondegenerate trilinear recurrence 2 with $A=a$ and $B=b$

Note: If the integrand has a positive integer exponent, decrementing it, rather than another positive fractional exponent, produces simpler antiderivatives.

Rule 1.1.1.3.13: If $m>1\,\wedge\,m+n+p+1\neq 0\,\wedge\,m\in\mathbb{Z}$, then

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\,\longrightarrow$$
$$\frac{b\,(a+b\,x)^{m-1}\,(c+d\,x)^{n+1}\,(e+f\,x)^{p+1}}{d\,f\,(m+n+p+1)}+$$
$$\frac{1}{d\,f\,(m+n+p+1)}\int (a+b\,x)^{m-2}\,(c+d\,x)^n\,(e+f\,x)^p\,\cdot$$
$$\left(a^2\,d\,f\,(m+n+p+1)-b\,(b\,c\,e\,(m-1)+a\,(d\,e\,(n+1)+c\,f\,(p+1)))+b\,(a\,d\,f\,(2\,m+n+p)-b\,(d\,e\,(m+n)+c\,f\,(m+p)))\,x\right)dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  b*(a+b*x)^(m-1)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/(d*f*(m+n+p+1)) +
  1/(d*f*(m+n+p+1))*Int[(a+b*x)^(m-2)*(c+d*x)^n*(e+f*x)^p*
    Simp[a^2*d*f*(m+n+p+1)-b*(b*c*e*(m-1)+a*(d*e*(n+1)+c*f*(p+1)))+b*(a*d*f*(2*m+n+p)-b*(d*e*(m+n)+c*f*(m+p)))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && GtQ[m,1] && NeQ[m+n+p+1,0] && IntegerQ[m]
```

14: $\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx$ when $m>0 \wedge n>0 \wedge m+n+p+1\neq 0$

Derivation: Nondegenerate trilinear recurrence 2 with $A = c$ and $B = d$

Rule 1.1.1.3.14: If $m>0 \wedge n>0 \wedge m+n+p+1\neq 0$, then

$$\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx \longrightarrow$$

$$\frac{(a+b x)^m (c+d x)^n (e+f x)^{p+1}}{f (m+n+p+1)} -$$

$$\frac{1}{f (m+n+p+1)} \int (a+b x)^{m-1} (c+d x)^{n-1} (e+f x)^p (c m (b e-a f)+a n (d e-c f)+(d m (b e-a f)+b n (d e-c f)) x) \, dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  (a+b*x)^m*(c+d*x)^n*(e+f*x)^(p+1)/(f*(m+n+p+1)) -
  1/(f*(m+n+p+1))*Int[(a+b*x)^(m-1)*(c+d*x)^(n-1)*(e+f*x)^p*
    Simp[c*m*(b*e-a*f)+a*n*(d*e-c*f)+(d*m*(b*e-a*f)+b*n*(d*e-c*f))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,p},x] && GtQ[m,0] && GtQ[n,0] && NeQ[m+n+p+1,0] && (IntegersQ[2*m,2*n,2*p] || (IntegersQ[m,n+p] || IntegersQ[p,m
```

15: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m>1\,\land\,m+n+p+1\neq 0$

Derivation: Nondegenerate trilinear recurrence 2 with $A=a$ and $B=b$

Rule 1.1.1.3.15: If $m>1\,\land\,m+n+p+1\neq 0$, then

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\,\longrightarrow$$

$$\frac{b\,(a+b\,x)^{m-1}\,(c+d\,x)^{n+1}\,(e+f\,x)^{p+1}}{d\,f\,(m+n+p+1)}+$$

$$\frac{1}{d\,f\,(m+n+p+1)}\int (a+b\,x)^{m-2}\,(c+d\,x)^n\,(e+f\,x)^p\,\cdot$$

$$\left(a^2\,d\,f\,(m+n+p+1)-b\,(b\,c\,e\,(m-1)+a\,(d\,e\,(n+1)+c\,f\,(p+1)))+b\,(a\,d\,f\,(2\,m+n+p)-b\,(d\,e\,(m+n)+c\,f\,(m+p)))\,x\right)dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  b*(a+b*x)^(m-1)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/(d*f*(m+n+p+1)) +
  1/(d*f*(m+n+p+1))*Int[(a+b*x)^(m-2)*(c+d*x)^n*(e+f*x)^p*
    Simp[a^2*d*f*(m+n+p+1)-b*(b*c*e*(m-1)+a*(d*e*(n+1)+c*f*(p+1)))+b*(a*d*f*(2*m+n+p)-b*(d*e*(m+n)+c*f*(m+p)))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && GtQ[m,1] && NeQ[m+n+p+1,0] && IntegersQ[2*m,2*n,2*p]
```

16: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m<-1$

Derivation: Nondegenerate trilinear recurrence 3 with $A=1$ and $B=0$

Note: If the integrand has a negative integer exponent, incrementing it, rather than another negative fractional exponent, produces simpler antiderivatives.

Rule 1.1.1.3.16: If $m<-1$, then

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx \longrightarrow$$

$$\frac{b\,(a+b\,x)^{m+1}\,(c+d\,x)^{n+1}\,(e+f\,x)^{p+1}}{(m+1)\,(b\,c-a\,d)\,(b\,e-a\,f)}+$$

$$\frac{1}{(m+1)\,(b\,c-a\,d)\,(b\,e-a\,f)}\int (a+b\,x)^{m+1}\,(c+d\,x)^n\,(e+f\,x)^p\,(a\,d\,f\,(m+1)-b\,(d\,e\,(m+n+2)+c\,f\,(m+p+2))-b\,d\,f\,(m+n+p+3)\,x)\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  b*(a+b*x)^(m+1)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/((m+1)*(b*c-a*d)*(b*e-a*f)) +
  1/((m+1)*(b*c-a*d)*(b*e-a*f))*Int[(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p*
    Simp[a*d*f*(m+1)-b*(d*e*(m+n+2)+c*f*(m+p+2))-b*d*f*(m+n+p+3)*x,x],x] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && ILtQ[m,-1] && (IntegerQ[n] || IntegersQ[2*n,2*p] || ILtQ[m+n+p+3,0])
```

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  b*(a+b*x)^(m+1)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/((m+1)*(b*c-a*d)*(b*e-a*f)) +
  1/((m+1)*(b*c-a*d)*(b*e-a*f))*Int[(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p*
    Simp[a*d*f*(m+1)-b*(d*e*(m+n+2)+c*f*(m+p+2))-b*d*f*(m+n+p+3)*x,x],x] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && LtQ[m,-1] && IntegersQ[2*m,2*n,2*p]
```

17. $\int \frac{(e+f x)^p}{(a+b x)\sqrt{c+d x}}\,dx$

1. $\int \frac{1}{(a+b x)\sqrt{c+d x}\,(e+f x)^{1/4}}\,dx$

1: $\int \frac{1}{(a+b x)\sqrt{c+d x}\,(e+f x)^{1/4}}\,dx$ when $-\frac{f}{d e-c f}>0$

Derivation: Integration by substitution

Basis: $\frac{1}{(a+b x)\sqrt{c+d x}\,(e+f x)^{1/4}} == -4\,\text{Subst}\left[\frac{x^2}{\left(b e-a f-b x^4\right)\sqrt{c-\frac{d e}{f}+\frac{d x^4}{f}}},\,x,\,(e+f x)^{1/4}\right]\partial_x (e+f x)^{1/4}$

Rule 1.1.1.3.17.1.1: If $-\frac{f}{d e-c f}>0$**, then**

$$\int \frac{1}{(a+b x)\sqrt{c+d x}\,(e+f x)^{1/4}}\,dx \longrightarrow -4\,\text{Subst}\left[\int \frac{x^2}{\left(b e-a f-b x^4\right)\sqrt{c-\frac{d e}{f}+\frac{d x^4}{f}}}\,dx,\,x,\,(e+f x)^{1/4}\right]$$

Program code:

```
Int[1/((a_.+b_.*x_)*Sqrt[c_.+d_.*x_]*(e_.+f_.*x_)^(1/4)),x_Symbol] :=
  -4*Subst[Int[x^2/((b*e-a*f-b*x^4)*Sqrt[c-d*e/f+d*x^4/f]),x],x,(e+f*x)^(1/4)] /;
FreeQ[{a,b,c,d,e,f},x] && GtQ[-f/(d*e-c*f),0]
```

2: $\int \frac{1}{(a+bx)\sqrt{c+dx}\,(e+fx)^{1/4}}\,dx$ when $-\frac{f}{de-cf} \not> 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{-\frac{f(c+dx)}{de-cf}}}{\sqrt{c+dx}} = 0$

- **Rule 1.1.1.3.17.1.2: If** $-\frac{f}{de-cf} \not> 0$**, then**

$$\int \frac{1}{(a+bx)\sqrt{c+dx}\,(e+fx)^{1/4}}\,dx \to \frac{\sqrt{-\frac{f(c+dx)}{de-cf}}}{\sqrt{c+dx}}\int \frac{1}{(a+bx)\sqrt{-\frac{cf}{de-cf}-\frac{dfx}{de-cf}}\,(e+fx)^{1/4}}\,dx$$

Program code:

```
Int[1/((a_.+b_.*x_)*Sqrt[c_.+d_.*x_]*(e_.+f_.*x_)^(1/4)),x_Symbol] :=
  Sqrt[-f*(c+d*x)/(d*e-c*f)]/Sqrt[c+d*x]*Int[1/((a+b*x)*Sqrt[-c*f/(d*e-c*f)-d*f*x/(d*e-c*f)]*(e+f*x)^(1/4)),x] /;
FreeQ[{a,b,c,d,e,f},x] && Not[GtQ[-f/(d*e-c*f),0]]
```

2. $\int \frac{1}{(a+b x) \sqrt{c+d x} (e+f x)^{3/4}} \, dx$

1: $\int \frac{1}{(a+b x) \sqrt{c+d x} (e+f x)^{3/4}} \, dx$ when $-\frac{f}{d e-c f} > 0$

Derivation: Integration by substitution

Basis: $\frac{1}{(a+b x) \sqrt{c+d x} (e+f x)^{3/4}} = -4 \text{Subst}\left[\frac{1}{(b e-a f-b x^4) \sqrt{c-\frac{d e}{f}+\frac{d x^4}{f}}}, x, (e+f x)^{1/4}\right] \partial_x (e+f x)^{1/4}$

Rule 1.1.1.3.17.2.1: If $-\frac{f}{d e-c f} > 0$**, then**

$$\int \frac{1}{(a+b x) \sqrt{c+d x} (e+f x)^{3/4}} \, dx \rightarrow -4 \text{Subst}\left[\int \frac{1}{\left(b e-a f-b x^4\right) \sqrt{c-\frac{d e}{f}+\frac{d x^4}{f}}} \, dx, x, (e+f x)^{1/4}\right]$$

Program code:

```
Int[1/((a_.+b_.*x_)*Sqrt[c_.+d_.*x_]*(e_.+f_.*x_)^(3/4)),x_Symbol] :=
  -4*Subst[Int[1/((b*e-a*f-b*x^4)*Sqrt[c-d*e/f+d*x^4/f]),x],x,(e+f*x)^(1/4)] /;
FreeQ[{a,b,c,d,e,f},x] && GtQ[-f/(d*e-c*f),0]
```

2: $\int \frac{1}{(a+b x)\sqrt{c+d x}\,(e+f x)^{3/4}}\,dx$ when $-\frac{f}{d e-c f} \not> 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{-\frac{f (c+d x)}{d e-c f}}}{\sqrt{c+d x}} == 0$

- **Rule 1.1.1.3.17.2.2: If** $-\frac{f}{d e-c f} \not> 0$**, then**

$$\int \frac{1}{(a+b x)\sqrt{c+d x}\,(e+f x)^{3/4}}\,dx \to \frac{\sqrt{-\frac{f (c+d x)}{d e-c f}}}{\sqrt{c+d x}} \int \frac{1}{(a+b x)\sqrt{-\frac{c f}{d e-c f}-\frac{d f x}{d e-c f}}\,(e+f x)^{3/4}}\,dx$$

Program code:

```
Int[1/((a_.+b_.*x_)*Sqrt[c_.+d_.*x_]*(e_.+f_.*x_)^(3/4)),x_Symbol] :=
  Sqrt[-f*(c+d*x)/(d*e-c*f)]/Sqrt[c+d*x]*Int[1/((a+b*x)*Sqrt[-c*f/(d*e-c*f)-d*f*x/(d*e-c*f)]*(e+f*x)^(3/4)),x] /;
FreeQ[{a,b,c,d,e,f},x] && Not[GtQ[-f/(d*e-c*f),0]]
```

18. $\int \frac{\sqrt{e+fx}}{\sqrt{a+bx}\,\sqrt{c+dx}}\,dx$

1. $\int \frac{\sqrt{e+fx}}{\sqrt{bx}\,\sqrt{c+dx}}\,dx$ when $de-cf\neq 0$

1. $\int \frac{\sqrt{e+fx}}{\sqrt{bx}\,\sqrt{c+dx}}\,dx$ when $de-cf\neq 0\land c>0\land e>0$

1: $\int \frac{\sqrt{e+fx}}{\sqrt{bx}\,\sqrt{c+dx}}\,dx$ when $de-cf\neq 0\bigwedge c>0\bigwedge e>0\bigwedge -\frac{b}{d}\nless 0$

- **Rule 1.1.1.3.18.1.1.1: If** $de-cf\neq 0\bigwedge c>0\bigwedge e>0\bigwedge -\frac{b}{d}>0$, **then**

$$\int \frac{\sqrt{e+fx}}{\sqrt{bx}\,\sqrt{c+dx}}\,dx \longrightarrow \frac{2\sqrt{e}}{b}\sqrt{-\frac{b}{d}}\,\text{EllipticE}\left[\text{ArcSin}\left[\frac{\sqrt{bx}}{\sqrt{c}\,\sqrt{-\frac{b}{d}}}\right],\frac{cf}{de}\right]$$

Program code:

```
Int[Sqrt[e_+f_.*x_]/(Sqrt[b_.*x_]*Sqrt[c_+d_.*x_]),x_Symbol] :=
  2*Sqrt[e]/b*Rt[-b/d,2]*EllipticE[ArcSin[Sqrt[b*x]/(Sqrt[c]*Rt[-b/d,2])],c*f/(d*e)] /;
FreeQ[{b,c,d,e,f},x] && NeQ[d*e-c*f,0] && GtQ[c,0] && GtQ[e,0] && Not[LtQ[-b/d,0]]
```

2: $\int \frac{\sqrt{e+f x}}{\sqrt{b x}\ \sqrt{c+d x}}\, dx$ when $d e-c f \neq 0 \bigwedge c>0 \bigwedge e>0 \bigwedge -\frac{b}{d}<0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{-F[x]}}{\sqrt{F[x]}} == 0$
- **Rule 1.1.1.3.18.1.1.2: If** $d e-c f \neq 0 \bigwedge c>0 \bigwedge e>0 \bigwedge -\frac{b}{d} \not> 0$**, then**

$$\int \frac{\sqrt{e+f x}}{\sqrt{b x}\ \sqrt{c+d x}}\, dx \longrightarrow \frac{\sqrt{-b x}}{\sqrt{b x}} \int \frac{\sqrt{e+f x}}{\sqrt{-b x}\ \sqrt{c+d x}}\, dx$$

Program code:

```
Int[Sqrt[e_+f_.*x_]/(Sqrt[b_.*x_]*Sqrt[c_+d_.*x_]),x_Symbol] :=
  Sqrt[-b*x]/Sqrt[b*x]*Int[Sqrt[e+f*x]/(Sqrt[-b*x]*Sqrt[c+d*x]),x] /;
FreeQ[{b,c,d,e,f},x] && NeQ[d*e-c*f,0] && GtQ[c,0] && GtQ[e,0] && LtQ[-b/d,0]
```

2: $\int \frac{\sqrt{e+f x}}{\sqrt{b x}\ \sqrt{c+d x}}\, dx$ when $d e-c f \neq 0 \wedge \neg (c>0 \wedge e>0)$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{e+f x}\ \sqrt{\frac{c+d x}{c}}}{\sqrt{c+d x}\ \sqrt{\frac{e+f x}{e}}} == 0$
- **Rule 1.1.1.3.18.1.2: If** $d e-c f \neq 0 \bigwedge \neg (c>0 \wedge e>0)$**, then**

$$\int \frac{\sqrt{e+f x}}{\sqrt{b x}\ \sqrt{c+d x}}\, dx \longrightarrow \frac{\sqrt{e+f x}\ \sqrt{1+\frac{d x}{c}}}{\sqrt{c+d x}\ \sqrt{1+\frac{f x}{e}}} \int \frac{\sqrt{1+\frac{f x}{e}}}{\sqrt{b x}\ \sqrt{1+\frac{d x}{c}}}\, dx$$

Program code:

```
Int[Sqrt[e_+f_.*x_]/(Sqrt[b_.*x_]*Sqrt[c_+d_.*x_]),x_Symbol] :=
  Sqrt[e+f*x]*Sqrt[1+d*x/c]/(Sqrt[c+d*x]*Sqrt[1+f*x/e])*Int[Sqrt[1+f*x/e]/(Sqrt[b*x]*Sqrt[1+d*x/c]),x] /;
FreeQ[{b,c,d,e,f},x] && NeQ[d*e-c*f,0] && Not[GtQ[c,0] && GtQ[e,0]]
```

2. $\int \frac{\sqrt{e+f x}}{\sqrt{a+b x}\,\sqrt{c+d x}}\,dx$

x: $\int \frac{\sqrt{e+f x}}{\sqrt{a+b x}\,\sqrt{c+d x}}\,dx$ when $b e == f(a-1)$

Derivation: Algebraic expansion

- **Basis: If** $b e == f(a-1)$**, then** $\frac{\sqrt{e+f x}}{\sqrt{a+b x}} == \frac{f\sqrt{a+b x}}{b\sqrt{e+f x}} - \frac{f}{b\sqrt{a+b x}\,\sqrt{e+f x}}$

Note: Instead of a single elliptic integral term, this rule produces two simpler such terms.

Rule 1.1.1.3.18.2.x: If $b e == f(a-1)$**, then**

$$\int \frac{\sqrt{e+f x}}{\sqrt{a+b x}\,\sqrt{c+d x}}\,dx \to \frac{f}{b}\int \frac{\sqrt{a+b x}}{\sqrt{c+d x}\,\sqrt{e+f x}}\,dx - \frac{f}{b}\int \frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}\,\sqrt{e+f x}}\,dx$$

Program code:

```
(* Int[Sqrt[e_.+f_.*x_]/(Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]),x_Symbol] :=
  f/b*Int[Sqrt[a+b*x]/(Sqrt[c+d*x]*Sqrt[e+f*x]),x] -
  f/b*Int[1/(Sqrt[a+b*x]*Sqrt[c+d*x]*Sqrt[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[b*e-f*(a-1),0] *)
```

x: $\int \frac{\sqrt{e+f x}}{\sqrt{a+b x}\,\sqrt{c+d x}}\,dx$ when $\frac{b}{b c-a d}>0\,\wedge\,\frac{b}{b e-a f}>0\,\wedge\,-\frac{b c-a d}{d}\not<0$

- **Rule 1.1.1.3.18.2.x: If** $\frac{b}{b c-a d}>0\,\wedge\,\frac{b}{b e-a f}>0\,\wedge\,-\frac{b c-a d}{d}\not<0$, **then**

$$\int \frac{\sqrt{e+f x}}{\sqrt{a+b x}\,\sqrt{c+d x}}\,dx \to \frac{2}{b}\sqrt{-\frac{b c-a d}{d}}\,\sqrt{\frac{b e-a f}{b c-a d}}\,\text{EllipticE}\left[\text{ArcSin}\left[\frac{\sqrt{a+b x}}{\sqrt{-\frac{b c-a d}{d}}}\right],\frac{f\,(b c-a d)}{d\,(b e-a f)}\right]$$

Program code:

```
(* Int[Sqrt[e_.+f_.*x_]/(Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]),x_Symbol] :=
  2/b*Rt[-(b*c-a*d)/d,2]*Sqrt[(b*e-a*f)/(b*c-a*d)]*
    EllipticE[ArcSin[Sqrt[a+b*x]/Rt[-(b*c-a*d)/d,2]],f*(b*c-a*d)/(d*(b*e-a*f))] /;
FreeQ[{a,b,c,d,e,f},x] && GtQ[b/(b*c-a*d),0] && GtQ[b/(b*e-a*f),0] && Not[LtQ[-(b*c-a*d)/d,0]] &&
  Not[SimplerQ[c+d*x,a+b*x] && GtQ[-d/(b*c-a*d),0] && GtQ[d/(d*e-c*f),0] && Not[LtQ[(b*c-a*d)/b,0]]] *)
```

1: $\int \frac{\sqrt{e+f x}}{\sqrt{a+b x}\,\sqrt{c+d x}}\,dx$ when $\frac{b}{b c-a d}>0\,\wedge\,\frac{b}{b e-a f}>0\,\wedge\,-\frac{b c-a d}{d}\not<0$

Derivation: Integration by substitution

- **Basis: If** $\frac{b}{b c-a d}>0\,\wedge\,\frac{b}{b e-a f}>0$, **then** $\frac{\sqrt{e+f x}}{\sqrt{a+b x}\,\sqrt{c+d x}} = \frac{2\sqrt{\frac{-b e+a f}{d}}}{b\sqrt{-\frac{b c-a d}{d}}}\,\text{Subst}\left[\frac{\sqrt{1+\frac{f x^2}{b e-a f}}}{\sqrt{1+\frac{d x^2}{b c-a d}}},x,\sqrt{a+b x}\right]\partial_x\sqrt{a+b x}$

- **Basis:** $\int \frac{\sqrt{1+\frac{f x^2}{b e-a f}}}{\sqrt{1+\frac{d x^2}{b c-a d}}}\,dx = \sqrt{-\frac{b c-a d}{d}}\,\text{EllipticE}\left[\text{ArcSin}\left[\frac{x}{\sqrt{-\frac{b c-a d}{d}}}\right],\frac{f\,(b c-a d)}{d\,(b e-a f)}\right]$

- **Rule 1.1.1.3.18.2.1: If** $\frac{b}{b c-a d}>0\,\wedge\,\frac{b}{b e-a f}>0\,\wedge\,-\frac{b c-a d}{d}\not<0$, **then**

$$\int \frac{\sqrt{e+f x}}{\sqrt{a+b x}\,\sqrt{c+d x}}\,dx \to \frac{2\sqrt{-\frac{b e-a f}{d}}}{b\sqrt{-\frac{b c-a d}{d}}}\,\text{Subst}\left[\int \frac{\sqrt{1+\frac{f x^2}{b e-a f}}}{\sqrt{1+\frac{d x^2}{b c-a d}}}\,dx,x,\sqrt{a+b x}\right]$$

$$\to \frac{2}{b}\sqrt{-\frac{b e-a f}{d}}\ \text{EllipticE}\left[\text{ArcSin}\left[\frac{\sqrt{a+b x}}{\sqrt{-\frac{b c-a d}{d}}}\right],\ \frac{f (b c-a d)}{d (b e-a f)}\right]$$

Program code:

```
Int[Sqrt[e_.+f_.*x_]/(Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]),x_Symbol] :=
  2/b*Rt[-(b*e-a*f)/d,2]*EllipticE[ArcSin[Sqrt[a+b*x]/Rt[-(b*c-a*d)/d,2]],f*(b*c-a*d)/(d*(b*e-a*f))] /;
FreeQ[{a,b,c,d,e,f},x] && GtQ[b/(b*c-a*d),0] && GtQ[b/(b*e-a*f),0] && Not[LtQ[-(b*c-a*d)/d,0]] &&
  Not[SimplerQ[c+d*x,a+b*x] && GtQ[-d/(b*c-a*d),0] && GtQ[d/(d*e-c*f),0] && Not[LtQ[(b*c-a*d)/b,0]]]
```

2: $\int \frac{\sqrt{e+f x}}{\sqrt{a+b x}\ \sqrt{c+d x}}\, dx$ when $\neg\left(\frac{b}{b c-a d}>0 \bigwedge \frac{b}{b e-a f}>0\right) \bigwedge -\frac{b c-a d}{d} \nless 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{e+f x}\ \sqrt{r (c+d x)}}{\sqrt{c+d x}\ \sqrt{s (e+f x)}} == 0$
- **Note:** $-\frac{b c-a d}{d} == \left(-\frac{b c-a d}{d} \ /.\ \left\{c \to \frac{b c}{b c-a d},\ d \to \frac{b d}{b c-a d},\ e \to \frac{b e}{b e-a f},\ f \to \frac{b f}{b e-a f}\right\}\right)$
- **Rule 1.1.1.3.18.2.2: If** $\neg\left(\frac{b}{b c-a d}>0 \bigwedge \frac{b}{b e-a f}>0\right) \bigwedge -\frac{b c-a d}{d} \nless 0$**, then**

$$\int \frac{\sqrt{e+f x}}{\sqrt{a+b x}\ \sqrt{c+d x}}\, dx \to \frac{\sqrt{e+f x}\ \sqrt{\frac{b (c+d x)}{b c-a d}}}{\sqrt{c+d x}\ \sqrt{\frac{b (e+f x)}{b e-a f}}} \int \frac{\sqrt{\frac{b e}{b e-a f}+\frac{b f x}{b e-a f}}}{\sqrt{a+b x}\ \sqrt{\frac{b c}{b c-a d}+\frac{b d x}{b c-a d}}}\, dx$$

Program code:

```
Int[Sqrt[e_.+f_.*x_]/(Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]),x_Symbol] :=
  Sqrt[e+f*x]*Sqrt[b*(c+d*x)/(b*c-a*d)]/(Sqrt[c+d*x]*Sqrt[b*(e+f*x)/(b*e-a*f)])*
    Int[Sqrt[b*e/(b*e-a*f)+b*f*x/(b*e-a*f)]/(Sqrt[a+b*x]*Sqrt[b*c/(b*c-a*d)+b*d*x/(b*c-a*d)]),x] /;
FreeQ[{a,b,c,d,e,f},x] && Not[GtQ[b/(b*c-a*d),0] && GtQ[b/(b*e-a*f),0]] && Not[LtQ[-(b*c-a*d)/d,0]]
```

19. $\int \frac{1}{\sqrt{a+b x}\ \sqrt{c+d x}\ \sqrt{e+f x}}\, dx$

1. $\int \frac{1}{\sqrt{b x} \sqrt{c + d x} \sqrt{e + f x}} \, dx$

1: $\int \frac{1}{\sqrt{b x} \sqrt{c + d x} \sqrt{e + f x}} \, dx$ when $c > 0 \land e > 0$

Rule 1.1.1.3.19.1.1: If $c > 0 \land e > 0$, then

$$\int \frac{1}{\sqrt{b x} \sqrt{c + d x} \sqrt{e + f x}} \, dx \to \frac{2}{b \sqrt{e}} \sqrt{-\frac{b}{d}} \, \text{EllipticF}\left[\text{ArcSin}\left[\frac{\sqrt{b x}}{\sqrt{c} \sqrt{-\frac{b}{d}}}\right], \frac{c f}{d e}\right]$$

Program code:

```
Int[1/(Sqrt[b_.*x_]*Sqrt[c_+d_.*x_]*Sqrt[e_+f_.*x_]),x_Symbol] :=
  2/(b*Sqrt[e])*Rt[-b/d,2]*EllipticF[ArcSin[Sqrt[b*x]/(Sqrt[c]*Rt[-b/d,2])],c*f/(d*e)] /;
FreeQ[{b,c,d,e,f},x] && GtQ[c,0] && GtQ[e,0] && (GtQ[-b/d,0] || LtQ[-b/f,0])
```

```
Int[1/(Sqrt[b_.*x_]*Sqrt[c_+d_.*x_]*Sqrt[e_+f_.*x_]),x_Symbol] :=
  2/(b*Sqrt[e])*Rt[-b/d,2]*EllipticF[ArcSin[Sqrt[b*x]/(Sqrt[c]*Rt[-b/d,2])],c*f/(d*e)] /;
FreeQ[{b,c,d,e,f},x] && GtQ[c,0] && GtQ[e,0] && (PosQ[-b/d] || NegQ[-b/f])
```

2: $\int \frac{1}{\sqrt{b x}\ \sqrt{c+d x}\ \sqrt{e+f x}}\, dx$ when $\neg\,(c>0 \land e>0)$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{1+\frac{d x}{c}}\ \sqrt{1+\frac{f x}{e}}}{\sqrt{c+d x}\ \sqrt{e+f x}} == 0$

- **Rule 1.1.1.3.19.1.2: If** $\neg\left(\frac{b}{b c-a d}>0 \bigwedge \frac{b}{b e-a f}>0\right)$**, then**

$$\int \frac{1}{\sqrt{b x}\ \sqrt{c+d x}\ \sqrt{e+f x}}\, dx \longrightarrow \frac{\sqrt{1+\frac{d x}{c}}\ \sqrt{1+\frac{f x}{e}}}{\sqrt{c+d x}\ \sqrt{e+f x}} \int \frac{1}{\sqrt{b x}\ \sqrt{1+\frac{d x}{c}}\ \sqrt{1+\frac{f x}{e}}}\, dx$$

Program code:

```
Int[1/(Sqrt[b_.*x_]*Sqrt[c_+d_.*x_]*Sqrt[e_+f_.*x_]),x_Symbol] :=
  Sqrt[1+d*x/c]*Sqrt[1+f*x/e]/(Sqrt[c+d*x]*Sqrt[e+f*x])*Int[1/(Sqrt[b*x]*Sqrt[1+d*x/c]*Sqrt[1+f*x/e]),x] /;
FreeQ[{b,c,d,e,f},x] && Not[GtQ[c,0] && GtQ[e,0]]
```

2. $\int \frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}\,\sqrt{e+f x}}\,dx$

1: $\int \frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}\,\sqrt{e+f x}}\,dx$ when $\frac{d}{b}>0\,\land\,\frac{f}{b}>0\,\land\,c\le\frac{a d}{b}\,\land\,e\le\frac{a f}{b}$

Derivation: Algebraic expansion and integration by substitution

- **Basis: If** $\frac{d}{b}>0\,\land\,c\le\frac{a d}{b}$**, then** $\sqrt{c+d x}=\sqrt{\frac{d}{b}}\,\sqrt{a+b x}\,\sqrt{\frac{b (c+d x)}{d (a+b x)}}$

- **Basis: If** $\frac{f}{b}>0\,\land\,e\le\frac{a f}{b}$**, then** $\sqrt{e+f x}=\sqrt{\frac{f}{b}}\,\sqrt{a+b x}\,\sqrt{\frac{b (e+f x)}{f (a+b x)}}$

- **Basis:** $\frac{\sqrt{\frac{b (c+d x)}{d (a+b x)}}}{\sqrt{a+b x}\,(c+d x)\,\sqrt{\frac{b (e+f x)}{f (a+b x)}}}=\frac{2}{d}\,\text{Subst}\Big[\frac{1}{x^2\,\sqrt{1+\frac{b c-a d}{d x^2}}\,\sqrt{1+\frac{b e-a f}{f x^2}}},\,x,\,\sqrt{a+b x}\Big]\,\partial_x\sqrt{a+b x}$

- **Basis:** $\int \frac{1}{x^2\,\sqrt{1+\frac{b c-a d}{d x^2}}\,\sqrt{1+\frac{b e-a f}{f x^2}}}\,dx=-\frac{1}{\sqrt{-\frac{b e-a f}{f}}}\,\text{EllipticF}\Big[\text{ArcSin}\Big[\frac{\sqrt{-\frac{b e-a f}{f}}}{x}\Big],\,\frac{f (b c-a d)}{d (b e-a f)}\Big]$

- **Rule 1.1.1.3.19.2.1: If** $\frac{d}{b}>0\,\land\,\frac{f}{b}>0\,\land\,c\le\frac{a d}{b}\,\land\,e\le\frac{a f}{b}$**, then**

$$\int \frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}\,\sqrt{e+f x}}\,dx\to\sqrt{\frac{d}{f}}\int \frac{\sqrt{\frac{b (c+d x)}{d (a+b x)}}}{\sqrt{a+b x}\,(c+d x)\,\sqrt{\frac{b (e+f x)}{f (a+b x)}}}\,dx$$

$$\to\frac{2\sqrt{\frac{d}{f}}}{d}\,\text{Subst}\Big[\int \frac{1}{x^2\,\sqrt{1+\frac{b c-a d}{d x^2}}\,\sqrt{1+\frac{b e-a f}{f x^2}}}\,dx,\,x,\,\sqrt{a+b x}\Big]$$

$$\rightarrow -\frac{2\sqrt{\frac{d}{f}}}{d\sqrt{-\frac{b e-a f}{f}}}\,\text{EllipticF}\left[\text{ArcSin}\left[\frac{\sqrt{-\frac{b e-a f}{f}}}{\sqrt{a+b x}}\right],\ \frac{f\,(b c-a d)}{d\,(b e-a f)}\right]$$

Program code:

```
Int[1/(Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]*Sqrt[e_+f_.*x_]),x_Symbol] :=
  -2*Sqrt[d/f]/(d*Rt[-(b*e-a*f)/f,2])*EllipticF[ArcSin[Rt[-(b*e-a*f)/f,2]/Sqrt[a+b*x]],f*(b*c-a*d)/(d*(b*e-a*f))] /;
FreeQ[{a,b,c,d,e,f},x] && GtQ[d/b,0] && GtQ[f/b,0] && LeQ[c,a*d/b] && LeQ[e,a*f/b]
```

x: $\int\frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}\,\sqrt{e+f x}}\,dx$ when $-\frac{b e-a f}{f}>0$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x\frac{\sqrt{c+d x}\,\sqrt{\frac{b\,(e+f x)}{f\,(a+b x)}}}{\sqrt{e+f x}\,\sqrt{\frac{b\,(c+d x)}{d\,(a+b x)}}}=0$

- **Basis:** $\frac{\sqrt{\frac{b\,(c+d x)}{d\,(a+b x)}}}{\sqrt{a+b x}\,(c+d x)\,\sqrt{\frac{b\,(e+f x)}{f\,(a+b x)}}}=\frac{2}{d}\,\text{Subst}\left[\frac{1}{x^2\sqrt{1+\frac{b c-a d}{d x^2}}\,\sqrt{1+\frac{b e-a f}{f x^2}}},\ x,\ \sqrt{a+b x}\right]\partial_x\sqrt{a+b x}$

- **Basis:** $\int\frac{1}{x^2\sqrt{1+\frac{b c-a d}{d x^2}}\,\sqrt{1+\frac{b e-a f}{f x^2}}}\,dx=-\frac{1}{\sqrt{-\frac{b e-a f}{f}}}\,\text{EllipticF}\left[\text{ArcSin}\left[\frac{\sqrt{-\frac{b e-a f}{f}}}{x}\right],\ \frac{f\,(b c-a d)}{d\,(b e-a f)}\right]$

- **Rule 1.1.1.3.19.2.1: If** $-\frac{b e-a f}{f}>0$**, then**

$$\int\frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}\,\sqrt{e+f x}}\,dx\rightarrow\frac{\sqrt{c+d x}\,\sqrt{\frac{b\,(e+f x)}{f\,(a+b x)}}}{\sqrt{e+f x}\,\sqrt{\frac{b\,(c+d x)}{d\,(a+b x)}}}\int\frac{\sqrt{\frac{b\,(c+d x)}{d\,(a+b x)}}}{\sqrt{a+b x}\,(c+d x)\,\sqrt{\frac{b\,(e+f x)}{f\,(a+b x)}}}\,dx$$

$$\rightarrow \frac{2\sqrt{c+d\,x}\,\sqrt{\frac{b\,(e+f\,x)}{f\,(a+b\,x)}}}{d\sqrt{e+f\,x}\,\sqrt{\frac{b\,(c+d\,x)}{d\,(a+b\,x)}}}\,\text{Subst}\left[\int\frac{1}{x^2\sqrt{1+\frac{b\,c-a\,d}{d\,x^2}}\,\sqrt{1+\frac{b\,e-a\,f}{f\,x^2}}}\,dx,\,x,\,\sqrt{a+b\,x}\right]$$

$$\rightarrow -\frac{2\sqrt{c+d\,x}\,\sqrt{\frac{b\,(e+f\,x)}{f\,(a+b\,x)}}}{d\sqrt{-\frac{b\,e-a\,f}{f}}\,\sqrt{e+f\,x}\,\sqrt{\frac{b\,(c+d\,x)}{d\,(a+b\,x)}}}\,\text{EllipticF}\left[\text{ArcSin}\left[\frac{\sqrt{-\frac{b\,e-a\,f}{f}}}{\sqrt{a+b\,x}}\right],\,\frac{f\,(b\,c-a\,d)}{d\,(b\,e-a\,f)}\right]$$

Program code:

```
(* Int[1/(Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]*Sqrt[e_+f_.*x_]),x_Symbol] :=
  -2*Sqrt[c+d*x]*Sqrt[b*(e+f*x)/(f*(a+b*x))]/(d*Rt[-(b*e-a*f)/f,2]*Sqrt[e+f*x]*Sqrt[b*(c+d*x)/(d*(a+b*x))])*
    EllipticF[ArcSin[Rt[-(b*e-a*f)/f,2]/Sqrt[a+b*x]],f*(b*c-a*d)/(d*(b*e-a*f))] /;
FreeQ[{a,b,c,d,e,f},x] && PosQ[-(b*e-a*f)/f] && (* (LtQ[-a/b,-c/d,-e/f] || GtQ[-a/b,-c/d,-e/f]) *)
  Not[SimplerQ[c+d*x,a+b*x] && (PosQ[(-d*e+c*f)/f] || PosQ[(b*e-a*f)/b])] &&
  Not[SimplerQ[e+f*x,a+b*x] && (PosQ[(b*e-a*f)/b] || PosQ[(b*c-a*d)/b])] *)
```

2: $\int\frac{1}{\sqrt{a+b\,x}\,\sqrt{c+d\,x}\,\sqrt{e+f\,x}}\,dx$ when $\frac{b\,c-a\,d}{b}>0\,\bigwedge\,\frac{b\,e-a\,f}{b}>0\,\bigwedge\,-\frac{b}{d}>0$

Derivation: Integration by substitution

- **Basis: If** $\frac{b\,c-a\,d}{b}>0\,\bigwedge\,\frac{b\,e-a\,f}{b}>0$, **then** $\frac{1}{\sqrt{a+b\,x}\,\sqrt{c+d\,x}\,\sqrt{e+f\,x}}==\frac{2}{b\sqrt{\frac{b\,c-a\,d}{b}}\,\sqrt{\frac{b\,e-a\,f}{b}}}\,\text{Subst}\left[\frac{1}{\sqrt{1+\frac{d\,x^2}{b\,c-a\,d}}\,\sqrt{1+\frac{f\,x^2}{b\,e-a\,f}}},\,x,\,\sqrt{a+b\,x}\right]\partial_x\sqrt{a+b\,x}$

- **Basis:** $\int\frac{1}{\sqrt{1+\frac{d\,x^2}{b\,c-a\,d}}\,\sqrt{1+\frac{f\,x^2}{b\,e-a\,f}}}\,dx==\sqrt{-\frac{b}{d}}\,\sqrt{\frac{b\,c-a\,d}{b}}\,\text{EllipticF}\left[\text{ArcSin}\left[\frac{x}{\sqrt{-\frac{b}{d}}\,\sqrt{\frac{b\,c-a\,d}{b}}}\right],\,\frac{f\,(b\,c-a\,d)}{d\,(b\,e-a\,f)}\right]$

- **Rule 1.1.1.3.19.2.2: If** $\frac{b\,c-a\,d}{b}>0\,\bigwedge\,\frac{b\,e-a\,f}{b}>0\,\bigwedge\,-\frac{b}{d}>0$, **then**

$$\int\frac{1}{\sqrt{a+b\,x}\,\sqrt{c+d\,x}\,\sqrt{e+f\,x}}\,dx\rightarrow\frac{2}{b\sqrt{\frac{b\,c-a\,d}{b}}\,\sqrt{\frac{b\,e-a\,f}{b}}}\,\text{Subst}\left[\int\frac{1}{\sqrt{1+\frac{d\,x^2}{b\,c-a\,d}}\,\sqrt{1+\frac{f\,x^2}{b\,e-a\,f}}}\,dx,\,x,\,\sqrt{a+b\,x}\right]$$

$$\to \frac{2\sqrt{-\frac{b}{d}}}{b\sqrt{\frac{b e-a f}{b}}}\,\text{EllipticF}\left[\text{ArcSin}\left[\frac{\sqrt{a+b x}}{\sqrt{-\frac{b}{d}}\,\sqrt{\frac{b c-a d}{b}}}\right],\ \frac{f\,(b c-a d)}{d\,(b e-a f)}\right]$$

Program code:

```
Int[1/(Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]*Sqrt[e_+f_.*x_]),x_Symbol] :=
  2*Rt[-b/d,2]/(b*Sqrt[(b*e-a*f)/b])*EllipticF[ArcSin[Sqrt[a+b*x]/(Rt[-b/d,2]*Sqrt[(b*c-a*d)/b])],f*(b*c-a*d)/(d*(b*e-a*f))] /;
FreeQ[{a,b,c,d,e,f},x] && GtQ[(b*c-a*d)/b,0] && GtQ[(b*e-a*f)/b,0] && PosQ[-b/d] &&
  Not[SimplerQ[c+d*x,a+b*x] && GtQ[(d*e-c*f)/d,0] && GtQ[-d/b,0]] &&
  Not[SimplerQ[c+d*x,a+b*x] && GtQ[(-b*e+a*f)/f,0] && GtQ[-f/b,0]] &&
  Not[SimplerQ[e+f*x,a+b*x] && GtQ[(-d*e+c*f)/f,0] && GtQ[(-b*e+a*f)/f,0] && (PosQ[-f/d] || PosQ[-f/b])]
```

```
Int[1/(Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]*Sqrt[e_+f_.*x_]),x_Symbol] :=
  2*Rt[-b/d,2]/(b*Sqrt[(b*e-a*f)/b])*EllipticF[ArcSin[Sqrt[a+b*x]/(Rt[-b/d,2]*Sqrt[(b*c-a*d)/b])],f*(b*c-a*d)/(d*(b*e-a*f))] /;
FreeQ[{a,b,c,d,e,f},x] && GtQ[b/(b*c-a*d),0] && GtQ[b/(b*e-a*f),0] && SimplerQ[a+b*x,c+d*x] && SimplerQ[a+b*x,e+f*x] &&
  (PosQ[-(b*c-a*d)/d] || NegQ[-(b*e-a*f)/f]) (* && PosQ[-b/d] *)
```

3: $\int \frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}\,\sqrt{e+f x}}\,dx$ when $\frac{b c-a d}{b} \ngtr 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{\frac{b\,(c+d\,x)}{b\,c-a\,d}}}{\sqrt{c+d\,x}} == 0$

- **Rule 1.1.1.3.19.2.3: If** $\frac{b c-a d}{b} \ngtr 0$**, then**

$$\int \frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}\,\sqrt{e+f x}}\,dx \to \frac{\sqrt{\frac{b\,(c+d\,x)}{b\,c-a\,d}}}{\sqrt{c+d x}}\int \frac{1}{\sqrt{a+b x}\,\sqrt{\frac{b c}{b c-a d}+\frac{b d x}{b c-a d}}\,\sqrt{e+f x}}\,dx$$

Program code:

```
Int[1/(Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]*Sqrt[e_+f_.*x_]),x_Symbol] :=
  Sqrt[b*(c+d*x)/(b*c-a*d)]/Sqrt[c+d*x]*Int[1/(Sqrt[a+b*x]*Sqrt[b*c/(b*c-a*d)+b*d*x/(b*c-a*d)]*Sqrt[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && Not[GtQ[(b*c-a*d)/b,0]] && SimplerQ[a+b*x,c+d*x] && SimplerQ[a+b*x,e+f*x]
```

4: $\int \frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}\,\sqrt{e+f x}}\,dx$ when $\frac{b e-a f}{b} \ngtr 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{\frac{b\,(e+f x)}{b e-a f}}}{\sqrt{e+f x}} == 0$
- **Rule 1.1.1.3.19.2.4: If** $\frac{b e-a f}{b} \ngtr 0$**, then**

$$\int \frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}\,\sqrt{e+f x}}\,dx \longrightarrow \frac{\sqrt{\frac{b\,(e+f x)}{b e-a f}}}{\sqrt{e+f x}} \int \frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}\,\sqrt{\frac{b e}{b e-a f}+\frac{b f x}{b e-a f}}}\,dx$$

- **Program code:**

```
Int[1/(Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]*Sqrt[e_+f_.*x_]),x_Symbol] :=
  Sqrt[b*(e+f*x)/(b*e-a*f)]/Sqrt[e+f*x]*Int[1/(Sqrt[a+b*x]*Sqrt[c+d*x]*Sqrt[b*e/(b*e-a*f)+b*f*x/(b*e-a*f)]),x] /;
FreeQ[{a,b,c,d,e,f},x] && Not[GtQ[(b*e-a*f)/b,0]]
```

20. $\int \frac{(a+b x)^m}{(c+d x)^{1/3} (e+f x)^{1/3}} \, dx$ when $2 b d e - b c f - a d f == 0 \land m \in \mathbb{Z}^-$

1: $\int \frac{1}{(a+b x) (c+d x)^{1/3} (e+f x)^{1/3}} \, dx$ when $2 b d e - b c f - a d f == 0$

- **Rule 1.1.1.3.20.1: If** $2 b d e - b c f - a d f == 0$**, let** $q = \left(\frac{b (b e-a f)}{(b c-a d)^2}\right)^{1/3}$**, then**

$$\int \frac{1}{(a+b x) (c+d x)^{1/3} (e+f x)^{1/3}} \, dx \longrightarrow -\frac{\text{Log}[a+b x]}{2 q (b c-a d)} - \frac{\sqrt{3} \, \text{ArcTan}\left[\frac{1}{\sqrt{3}} + \frac{2 q (c+d x)^{2/3}}{\sqrt{3} (e+f x)^{1/3}}\right]}{2 q (b c-a d)} + \frac{3 \, \text{Log}\left[q (c+d x)^{2/3} - (e+f x)^{1/3}\right]}{4 q (b c-a d)}$$

Program code:

```
Int[1/((a_.+b_.*x_)*(c_.+d_.*x_)^(1/3)*(e_.+f_.*x_)^(1/3)),x_Symbol] :=
  With[{q=Rt[b*(b*e-a*f)/(b*c-a*d)^2,3]},
  -Log[a+b*x]/(2*q*(b*c-a*d)) -
   Sqrt[3]*ArcTan[1/Sqrt[3]+2*q*(c+d*x)^(2/3)/(Sqrt[3]*(e+f*x)^(1/3))]/(2*q*(b*c-a*d)) +
   3*Log[q*(c+d*x)^(2/3)-(e+f*x)^(1/3)]/(4*q*(b*c-a*d))] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[2*b*d*e-b*c*f-a*d*f,0]
```

2: $\int \frac{(a+b x)^m}{(c+d x)^{1/3} (e+f x)^{1/3}} \, dx$ when $2 b d e - b c f - a d f == 0 \land m+1 \in \mathbb{Z}^-$

Derivation: Nondegenerate trilinear recurrence 3 with $A = 1$ **and** $B = 0$

Rule 1.1.1.3.20.2: If $2 b d e - b c f - a d f == 0 \land m+1 \in \mathbb{Z}^-$**, then**

$$\int \frac{(a+b x)^m}{(c+d x)^{1/3} (e+f x)^{1/3}} \, dx \longrightarrow$$

$$\frac{b (a+b x)^{m+1} (c+d x)^{2/3} (e+f x)^{2/3}}{(m+1) (b c-a d) (b e-a f)} + \frac{f}{6 (m+1) (b c-a d) (b e-a f)} \int \frac{(a+b x)^{m+1} (a d (3 m+1) - 3 b c (3 m+5) - 2 b d (3 m+7) x)}{(c+d x)^{1/3} (e+f x)^{1/3}} \, dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_/((c_.+d_.*x_)^(1/3)*(e_.+f_.*x_)^(1/3)),x_Symbol] :=
  b*(a+b*x)^(m+1)*(c+d*x)^(2/3)*(e+f*x)^(2/3)/((m+1)*(b*c-a*d)*(b*e-a*f)) +
  f/(6*(m+1)*(b*c-a*d)*(b*e-a*f))*
    Int[(a+b*x)^(m+1)*(a*d*(3*m+1)-3*b*c*(3*m+5)-2*b*d*(3*m+7)*x)/((c+d*x)^(1/3)*(e+f*x)^(1/3)),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[2*b*d*e-b*c*f-a*d*f,0] && ILtQ[m,-1]
```

21. $\int (a+b\,x)^m\,(c+d\,x)^n\,(f\,x)^p\,dx$ when $b\,c+a\,d==0\,\wedge\,m-n==0$

1: $\int (a+b\,x)^m\,(c+d\,x)^n\,(f\,x)^p\,dx$ when $b\,c+a\,d==0\,\wedge\,m-n==0\,\wedge\,a>0\,\wedge\,c>0$

Derivation: Algebraic simplification

Basis: If $b\,c+a\,d==0\,\wedge\,a>0\,\wedge\,c>0$**, then** $(a+b\,x)^m\,(c+d\,x)^m==\left(a\,c+b\,d\,x^2\right)^m$

Rule 1.1.1.3.21.1: If $b\,c+a\,d==0\,\wedge\,m-n==0\,\wedge\,a>0\,\wedge\,c>0$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(f\,x)^p\,dx\;\longrightarrow\;\int \left(a\,c+b\,d\,x^2\right)^m\,(f\,x)^p\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(f_.*x_)^p_.,x_Symbol] :=
  Int[(a*c+b*d*x^2)^m*(f*x)^p,x] /;
FreeQ[{a,b,c,d,f,m,n,p},x] && EqQ[b*c+a*d,0] && EqQ[m-n,0] && GtQ[a,0] && GtQ[c,0]
```

2: $\int (a+b\,x)^m\,(c+d\,x)^n\,(f\,x)^p\,dx$ when $b\,c+a\,d==0\,\wedge\,m-n==0$

Derivation: Piecewise constant extraction

- **Basis: If** $b\,c+a\,d==0$**, then** $\partial_x\frac{(a+b\,x)^m\,(c+d\,x)^m}{\left(a\,c+b\,d\,x^2\right)^m}==0$
- **Basis: If** $b\,c+a\,d==0$**, then** $\frac{(a+b\,x)^m\,(c+d\,x)^m}{\left(a\,c+b\,d\,x^2\right)^m}==\frac{(a+b\,x)^{\text{FracPart}[m]}\,(c+d\,x)^{\text{FracPart}[m]}}{\left(a\,c+b\,d\,x^2\right)^{\text{FracPart}[m]}}$

Rule 1.1.1.3.21.2: If $b\,c+a\,d==0\,\wedge\,m-n==0$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(f\,x)^p\,dx\;\longrightarrow\;\frac{(a+b\,x)^{\text{FracPart}[m]}\,(c+d\,x)^{\text{FracPart}[m]}}{\left(a\,c+b\,d\,x^2\right)^{\text{FracPart}[m]}}\int \left(a\,c+b\,d\,x^2\right)^m\,(f\,x)^p\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(f_.*x_)^p_.,x_Symbol] :=
  (a+b*x)^FracPart[m]*(c+d*x)^FracPart[m]/(a*c+b*d*x^2)^FracPart[m]*Int[(a*c+b*d*x^2)^m*(f*x)^p,x] /;
FreeQ[{a,b,c,d,f,m,n,p},x] && EqQ[b*c+a*d,0] && EqQ[m-n,0]
```

22: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m \in \mathbb{Z}^+ \lor (m \mid n) \in \mathbb{Z}^-$

Derivation: Algebraic expansion

Rule 1.1.1.3.22: If $m \in \mathbb{Z}^+ \lor (m \mid n) \in \mathbb{Z}^-$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx \longrightarrow \int \text{ExpandIntegrand}\left[(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p,\,x\right]dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x)^m*(c+d*x)^n*(e+f*x)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && (IGtQ[m,0] || ILtQ[m,0] && ILtQ[n,0])
```

23: $\int (e\,x)^p\,(a+b\,x)^m\,(c+d\,x)^n\,dx$ when $b\,c-a\,d \neq 0 \land p \in \mathbb{F} \land m \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$**, then** $(e\,x)^p\,F[x] = \frac{k}{e}\,\text{Subst}\left[x^{k\,(p+1)-1}\,F\left[\frac{x^k}{e}\right],\,x,\,(e\,x)^{1/k}\right]\,\partial_x\,(e\,x)^{1/k}$

Rule 1.1.1.3.23 If $b\,c-a\,d \neq 0 \land p \in \mathbb{F} \land m \in \mathbb{Z}$**, let** $k = \text{Denominator}[p]$**, then**

$$\int (e\,x)^p\,(a+b\,x)^m\,(c+d\,x)^n\,dx \longrightarrow \frac{k}{e}\,\text{Subst}\left[\int x^{k\,(p+1)-1}\left(a+\frac{b\,x^k}{e}\right)^m\left(c+\frac{d\,x^k}{e}\right)^n dx,\,x,\,(e\,x)^{1/k}\right]$$

Program code:

```
Int[(e_.*x_)^p_*(a_+b_.*x_)^m_*(c_+d_.*x_)^n_,x_Symbol] :=
  With[{k=Denominator[p]},
  k/e*Subst[Int[x^(k*(p+1)-1)*(a+b*x^k/e)^m*(c+d*x^k/e)^n,x],x,(e*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[b*c-a*d,0] && FractionQ[p] && IntegerQ[m]
```

24. $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m+n+p\in\mathbb{Z}$

1: $\int \frac{(a+b\,x)^m\,(c+d\,x)^n}{(e+f\,x)^2}\,dx$ when $m+n\in\mathbb{Z}^+\wedge 2\,b\,d\,e-f\,(b\,c+a\,d)==0$

Derivation: Algebraic expansion

Basis: $\frac{(a+b\,x)^m\,(c+d\,x)^n}{(e+f\,x)^2}==\frac{b\,d}{f^2}\,(a+b\,x)^{m-1}\,(c+d\,x)^{n-1}+\frac{(b\,e-a\,f)\,(d\,e-c\,f)\,(a+b\,x)^{m-1}\,(c+d\,x)^{n-1}}{f^2\,(e+f\,x)^2}$

Rule 1.1.1.3.24.1: If $m+n\in\mathbb{Z}^+\wedge 2\,b\,d\,e-f\,(b\,c+a\,d)==0$**, then**

$$\int \frac{(a+b\,x)^m\,(c+d\,x)^n}{(e+f\,x)^2}\,dx\,\longrightarrow\,\frac{b\,d}{f^2}\int (a+b\,x)^{m-1}\,(c+d\,x)^{n-1}\,dx+\frac{(b\,e-a\,f)\,(d\,e-c\,f)}{f^2}\int \frac{(a+b\,x)^{m-1}\,(c+d\,x)^{n-1}}{(e+f\,x)^2}\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_/(e_.+f_.*x_)^2,x_Symbol] :=
  b*d/f^2*Int[(a+b*x)^(m-1)*(c+d*x)^(n-1),x] +
  (b*e-a*f)*(d*e-c*f)/f^2*Int[(a+b*x)^(m-1)*(c+d*x)^(n-1)/(e+f*x)^2,x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && IGtQ[m+n,0] && EqQ[2*b*d*e-f*(b*c+a*d),0]
```

2: $\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx$ when $m+n+p=0 \land p \in \mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: If $m+n+p=0$, **then** $(a+b x)^m (c+d x)^n (e+f x)^p =$
$$\frac{f^{p-1} (a+b x)^m (d e p-c f (p-1)+d f x)}{d^p (c+d x)^{m+1}}+\frac{f^{p-1} (a+b x)^m (e+f x)^p}{(c+d x)^{m+1}} \left(f^{-p+1} (c+d x)^{-p+1}-d^{-p} (d e p-c f (p-1)+d f x) (e+f x)^{-p}\right)$$

Note: If $p \in \mathbb{Z}^-$, **then** $f^{-p+1} (c+d x)^{-p+1}-d^{-p} (d e p-c f (p-1)+d f x) (e+f x)^{-p}$ **is a polynomial of degree** $-p-1$ **in** x.

Rule 1.1.1.3.24.2: If $m+n+p=0 \land p \in \mathbb{Z}^-$, **then**

$$\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx \longrightarrow$$
$$\frac{f^{p-1}}{d^p} \int \frac{(a+b x)^m (d e p-c f (p-1)+d f x)}{(c+d x)^{m+1}} \, dx+f^{p-1} \int \frac{(a+b x)^m (e+f x)^p}{(c+d x)^{m+1}} \left(f^{-p+1} (c+d x)^{-p+1}-d^{-p} (d e p-c f (p-1)+d f x) (e+f x)^{-p}\right) dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_,x_Symbol] :=
  f^(p-1)/d^p*Int[(a+b*x)^m*(d*e*p-c*f*(p-1)+d*f*x)/(c+d*x)^(m+1),x] +
  f^(p-1)*Int[(a+b*x)^m*(e+f*x)^p/(c+d*x)^(m+1)*
    ExpandToSum[f^(-p+1)*(c+d*x)^(-p+1)-(d*e*p-c*f*(p-1)+d*f*x)/(d^p*(e+f*x)^p),x],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[m+n+p,0] && ILtQ[p,0] && (LtQ[m,0] || SumSimplerQ[m,1] || Not[LtQ[n,0] || SumSimplerQ[n,1]])
```

3: $\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx$ when $m+n+p+1=0 \land p \in \mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: If $m+n+p+1=0$**, then** $(a+b x)^m (c+d x)^n (e+f x)^p = \frac{b\, d^{m+n} f^p (a+b x)^{m-1}}{(c+d x)^m} + \frac{(a+b x)^{m-1} (e+f x)^p}{(c+d x)^m} \left((a+b x) (c+d x)^{-p-1} - b\, d^{-p-1} f^p (e+f x)^{-p}\right)$

Note: If $p \in \mathbb{Z}^-$**, then** $(a+b x) (c+d x)^{-p-1} - b\, d^{-p-1} f^p (e+f x)^{-p}$ **is a polynomial of degree** $-p-1$ **in x.**

Rule 1.1.1.3.24.3: If $m+n+p+1=0 \land p \in \mathbb{Z}^-$**, then**

$$\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx \longrightarrow b\, d^{m+n} f^p \int \frac{(a+b x)^{m-1}}{(c+d x)^m} \, dx + \int \frac{(a+b x)^{m-1} (e+f x)^p}{(c+d x)^m} \left((a+b x) (c+d x)^{-p-1} - b\, d^{-p-1} f^p (e+f x)^{-p}\right) dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_,x_Symbol] :=
  b*d^(m+n)*f^p*Int[(a+b*x)^(m-1)/(c+d*x)^m,x] +
  Int[(a+b*x)^(m-1)*(e+f*x)^p/(c+d*x)^m*ExpandToSum[(a+b*x)*(c+d*x)^(-p-1)-(b*d^(-p-1)*f^p)/(e+f*x)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[m+n+p+1,0] && ILtQ[p,0] && (GtQ[m,0] || SumSimplerQ[m,-1] || Not[GtQ[n,0] || SumSimplerQ[n,-1]])
```

4. $\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx$ when $m+n+p+2=0$

1: $\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx$ when $m+n+p+2=0 \land n \in \mathbb{Z}^-$

Rule 1.1.1.3.24.4.1: If $m+n+p+2=0 \land n \in \mathbb{Z}^-$**, then**

$$\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx \longrightarrow$$
$$\frac{(b c-a d)^n (a+b x)^{m+1}}{(m+1) (b e-a f)^{n+1} (e+f x)^{m+1}} \text{Hypergeometric2F1}\left[m+1, -n, m+2, -\frac{(d e-c f) (a+b x)}{(b c-a d) (e+f x)}\right]$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_,x_Symbol] :=
  (b*c-a*d)^n*(a+b*x)^(m+1)/((m+1)*(b*e-a*f)^(n+1)*(e+f*x)^(m+1))*
    Hypergeometric2F1[m+1,-n,m+2,-(d*e-c*f)*(a+b*x)/((b*c-a*d)*(e+f*x))] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && EqQ[m+n+p+2,0] && ILtQ[n,0] && (SumSimplerQ[m,1] || Not[SumSimplerQ[p,1]]) && Not[ILtQ[m,0]]
```

2: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m+n+p+2==0\,\wedge\,n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x\left(\frac{(c+d\,x)^n}{(e+f\,x)^n}\left(\frac{(b\,e-a\,f)\,(c+d\,x)}{(b\,c-a\,d)\,(e+f\,x)}\right)^{-n}\right)==0$

Rule 1.1.1.3.24.4.2: If $m+n+p+2==0\,\wedge\,n\notin\mathbb{Z}$, then

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$$

$$\longrightarrow\ \frac{(c+d\,x)^n}{(e+f\,x)^n}\left(\frac{(b\,e-a\,f)\,(c+d\,x)}{(b\,c-a\,d)\,(e+f\,x)}\right)^{-n}\int\frac{(a+b\,x)^m}{(e+f\,x)^{m+2}}\left(\frac{(b\,e-a\,f)\,(c+d\,x)}{(b\,c-a\,d)\,(e+f\,x)}\right)^n dx$$

$$\longrightarrow\ \frac{(a+b\,x)^{m+1}\,(c+d\,x)^n\,(e+f\,x)^{p+1}}{(b\,e-a\,f)\,(m+1)}\left(\frac{(b\,e-a\,f)\,(c+d\,x)}{(b\,c-a\,d)\,(e+f\,x)}\right)^{-n}\text{Hypergeometric2F1}\left[m+1,\,-n,\,m+2,\,-\frac{(d\,e-c\,f)\,(a+b\,x)}{(b\,c-a\,d)\,(e+f\,x)}\right]$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^(p+1)/((b*e-a*f)*(m+1))*((b*e-a*f)*(c+d*x)/((b*c-a*d)*(e+f*x)))^(-n)*
    Hypergeometric2F1[m+1,-n,m+2,-(d*e-c*f)*(a+b*x)/((b*c-a*d)*(e+f*x))] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && EqQ[m+n+p+2,0] && Not[IntegerQ[n]]
```

5: $\int \frac{(a+b x)^m (c+d x)^n}{e+f x} \, dx$ when $m+n+1 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

- **Basis:** $\frac{(a+b x)^m (c+d x)^n}{e+f x} = \frac{(c f-d e)^{m+n+1} (a+b x)^m}{f^{m+n+1} (c+d x)^{m+1} (e+f x)} + \frac{(a+b x)^m}{f^{m+n+1} (c+d x)^{m+1}} \frac{f^{m+n+1} (c+d x)^{m+n+1}-(c f-d e)^{m+n+1}}{e+f x}$
- **Note: If** $m+n+1 \in \mathbb{Z}^+$**, then** $\frac{f^{m+n+1} (c+d x)^{m+n+1}-(c f-d e)^{m+n+1}}{e+f x}$ **is a polynomial in** x**.**

Rule 1.1.1.3.24.5: If $m+n+1 \in \mathbb{Z}^+$**, then**

$$\int \frac{(a+b x)^m (c+d x)^n}{e+f x} \, dx \longrightarrow \frac{(c f-d e)^{m+n+1}}{f^{m+n+1}} \int \frac{(a+b x)^m}{(c+d x)^{m+1} (e+f x)} \, dx + \frac{1}{f^{m+n+1}} \int \frac{(a+b x)^m}{(c+d x)^{m+1}} \frac{f^{m+n+1} (c+d x)^{m+n+1}-(c f-d e)^{m+n+1}}{e+f x} \, dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_/(e_.+f_.*x_),x_Symbol] :=
  (c*f-d*e)^(m+n+1)/f^(m+n+1)*Int[(a+b*x)^m/((c+d*x)^(m+1)*(e+f*x)),x] +
  1/f^(m+n+1)*Int[(a+b*x)^m/(c+d*x)^(m+1)*ExpandToSum[(f^(m+n+1)*(c+d*x)^(m+n+1)-(c*f-d*e)^(m+n+1))/(e+f*x),x],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && IGtQ[m+n+1,0] && (LtQ[m,0] || SumSimplerQ[m,1] || Not[LtQ[n,0] || SumSimplerQ[n,1]])
```

6: $\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx$ when $m+n+p+2 \in \mathbb{Z}^- \wedge m \neq -1$

Derivation: Nondegenerate trilinear recurrence 3 with $A = 1$ and $B = 0$

Note: If $m+n+p+2 \in \mathbb{Z}^-$, then $\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx$ can be expressed in terms of the hypergeometric function 2F1.

Rule 1.1.1.3.24.6: If $m+n+p+2 \in \mathbb{Z}^- \wedge m \neq -1$, then

$$\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx \longrightarrow$$
$$\frac{b (a+b x)^{m+1} (c+d x)^{n+1} (e+f x)^{p+1}}{(m+1) (b c-a d) (b e-a f)} +$$
$$\frac{1}{(m+1) (b c-a d) (b e-a f)} \int (a+b x)^{m+1} (c+d x)^n (e+f x)^p (a d f (m+1) - b (d e (m+n+2) + c f (m+p+2)) - b d f (m+n+p+3) x) \, dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  b*(a+b*x)^(m+1)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/((m+1)*(b*c-a*d)*(b*e-a*f)) +
  1/((m+1)*(b*c-a*d)*(b*e-a*f))*Int[(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p*
    Simp[a*d*f*(m+1)-b*(d*e*(m+n+2)+c*f*(m+p+2))-b*d*f*(m+n+p+3)*x,x],x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && ILtQ[m+n+p+2,0] && NeQ[m,-1] && (SumSimplerQ[m,1] || Not[SumSimplerQ[n,1]] && Not[SumSimplerQ[p,1]]
```

25: $\int (a+b x)^m (c+d x)^n (f x)^p \, dx$ when $b c + a d == 0 \wedge m-n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Note: Integrals of this form can be expressed in terms of the confluent hypergeometric function $2 F 1$ instead of requiring the Appell hypergeometric function.

Rule 1.1.1.3.25: If $b c + a d == 0 \wedge m-n \in \mathbb{Z}^+$, then

$$\int (a+b x)^m (c+d x)^n (f x)^p \, dx \longrightarrow \int (a+b x)^n (c+d x)^n (f x)^p \, \text{ExpandIntegrand}[(a+b x)^{m-n}, x] \, dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(f_.*x_)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x)^n*(c+d*x)^n*(f*x)^p,(a+b*x)^(m-n),x],x] /;
FreeQ[{a,b,c,d,f,m,n,p},x] && EqQ[b*c+a*d,0] && IGtQ[m-n,0] && NeQ[m+n+p+2,0]
```

A. $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m\notin\mathbb{Z}\land n\notin\mathbb{Z}$

1. $\int (b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m\notin\mathbb{Z}\land n\notin\mathbb{Z}$

1: $\int (b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m\notin\mathbb{Z}\land n\notin\mathbb{Z}\land c>0\land(p\in\mathbb{Z}\lor e>0)$

Rule 1.1.1.3.A.1.1: If $m\notin\mathbb{Z}\land n\notin\mathbb{Z}\land c>0\land(p\in\mathbb{Z}\lor e>0)$, **then**

$$\int (b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\,\longrightarrow\,\frac{c^n\,e^p\,(b\,x)^{m+1}}{b\,(m+1)}\,\text{AppellF1}\left[m+1,\,-n,\,-p,\,m+2,\,-\frac{d\,x}{c},\,-\frac{f\,x}{e}\right]$$

Program code:

```
Int[(b_.*x_)^m_*(c_+d_.*x_)^n_*(e_+f_.*x_)^p_,x_Symbol] :=
  c^n*e^p*(b*x)^(m+1)/(b*(m+1))*AppellF1[m+1,-n,-p,m+2,-d*x/c,-f*x/e] /;
FreeQ[{b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && GtQ[c,0] && (IntegerQ[p] || GtQ[e,0])
```

2: $\int (b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m\notin\mathbb{Z}\land n\notin\mathbb{Z}\land -\frac{d}{b\,c}>0\land\left(p\in\mathbb{Z}\lor\frac{d}{d\,e-c\,f}>0\right)$

Rule 1.1.1.3.A.1.2: If $m\notin\mathbb{Z}\land n\notin\mathbb{Z}\land -\frac{d}{b\,c}>0\land\left(p\in\mathbb{Z}\lor\frac{d}{d\,e-c\,f}>0\right)$, **then**

$$\int (b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\,\longrightarrow$$

$$\frac{(c+d\,x)^{n+1}}{d\,(n+1)\,\left(-\frac{d}{b\,c}\right)^m\,\left(\frac{d}{d\,e-c\,f}\right)^p}\,\text{AppellF1}\left[n+1,\,-m,\,-p,\,n+2,\,1+\frac{d\,x}{c},\,-\frac{f\,(c+d\,x)}{d\,e-c\,f}\right]$$

Program code:

```
Int[(b_.*x_)^m_*(c_+d_.*x_)^n_*(e_+f_.*x_)^p_,x_Symbol] :=
  (c+d*x)^(n+1)/(d*(n+1)*(-d/(b*c))^m*(d/(d*e-c*f))^p)*AppellF1[n+1,-m,-p,n+2,1+d*x/c,-f*(c+d*x)/(d*e-c*f)] /;
FreeQ[{b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && GtQ[-d/(b*c),0] && (IntegerQ[p] || GtQ[d/(d*e-c*f),0])
```

3: $\int (b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m\notin\mathbb{Z}\,\wedge\,n\notin\mathbb{Z}\,\wedge\,c\ngtr 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(c+d\,x)^n}{\left(\frac{c+d\,x}{c}\right)^n} == 0$

Rule 1.1.1.3.A.1.3: If $m\notin\mathbb{Z}\,\wedge\,n\notin\mathbb{Z}\,\wedge\,c\ngtr 0$**, then**

$$\int (b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx \longrightarrow \frac{c^{\text{IntPart}[n]}\,(c+d\,x)^{\text{FracPart}[n]}}{\left(1+\frac{d\,x}{c}\right)^{\text{FracPart}[n]}}\int (b\,x)^m\left(1+\frac{d\,x}{c}\right)^n\,(e+f\,x)^p\,dx$$

Program code:

```
Int[(b_.*x_)^m_*(c_+d_.*x_)^n_*(e_+f_.*x_)^p_,x_Symbol] :=
  c^IntPart[n]*(c+d*x)^FracPart[n]/(1+d*x/c)^FracPart[n]*Int[(b*x)^m*(1+d*x/c)^n*(e+f*x)^p,x] /;
FreeQ[{b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && Not[GtQ[c,0]]
```

2. $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m\notin\mathbb{Z}\,\wedge\,n\notin\mathbb{Z}\,\wedge\,p\in\mathbb{Z}$

1: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m\notin\mathbb{Z}\,\bigwedge\,n\notin\mathbb{Z}\,\bigwedge\,p\in\mathbb{Z}\,\bigwedge\,\frac{b}{b\,c-a\,d}>0$

- **Rule 1.1.1.3.A.2.1: If** $m\notin\mathbb{Z}\,\bigwedge\,n\notin\mathbb{Z}\,\bigwedge\,p\in\mathbb{Z}\,\bigwedge\,\frac{b}{b\,c-a\,d}>0$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx \longrightarrow$$

$$\frac{(b\,e-a\,f)^p\,(a+b\,x)^{m+1}}{b^{p+1}\,(m+1)\,\left(\frac{b}{b\,c-a\,d}\right)^n}\,\text{AppellF1}\left[m+1,\,-n,\,-p,\,m+2,\,-\frac{d\,(a+b\,x)}{b\,c-a\,d},\,-\frac{f\,(a+b\,x)}{b\,e-a\,f}\right]$$

Program code:

```
Int[(a_+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_,x_Symbol] :=
  (b*e-a*f)^p*(a+b*x)^(m+1)/(b^(p+1)*(m+1)*(b/(b*c-a*d))^n)*
    AppellF1[m+1,-n,-p,m+2,-d*(a+b*x)/(b*c-a*d),-f*(a+b*x)/(b*e-a*f)] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && IntegerQ[p] && GtQ[b/(b*c-a*d),0] &&
  Not[GtQ[d/(d*a-c*b),0] && SimplerQ[c+d*x,a+b*x]]
```

2: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m\notin\mathbb{Z}\,\bigwedge\,n\notin\mathbb{Z}\,\bigwedge\,p\in\mathbb{Z}\,\bigwedge\,\frac{b}{b\,c-a\,d}\ngtr 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(c+d\,x)^n}{\left(\frac{b\,(c+d\,x)}{b\,c-a\,d}\right)^n} == 0$
- **Rule 1.1.1.3.A.2.2: If** $m\notin\mathbb{Z}\,\bigwedge\,n\notin\mathbb{Z}\,\bigwedge\,p\in\mathbb{Z}\,\bigwedge\,\frac{b}{b\,c-a\,d}\ngtr 0$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\,\longrightarrow\,\frac{(c+d\,x)^{\text{FracPart}[n]}}{\left(\frac{b}{b\,c-a\,d}\right)^{\text{IntPart}[n]}\left(\frac{b\,(c+d\,x)}{b\,c-a\,d}\right)^{\text{FracPart}[n]}}\int (a+b\,x)^m\left(\frac{b\,c}{b\,c-a\,d}+\frac{b\,d\,x}{b\,c-a\,d}\right)^n(e+f\,x)^p\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_,x_Symbol] :=
  (c+d*x)^FracPart[n]/((b/(b*c-a*d))^IntPart[n]*(b*(c+d*x)/(b*c-a*d))^FracPart[n])*
    Int[(a+b*x)^m*(b*c/(b*c-a*d)+b*d*x/(b*c-a*d))^n*(e+f*x)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && IntegerQ[p] && Not[GtQ[b/(b*c-a*d),0]] &&
  Not[SimplerQ[c+d*x,a+b*x]]
```

3. $\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx$ when $m \notin \mathbb{Z} \land n \notin \mathbb{Z} \land p \notin \mathbb{Z}$

1. $\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx$ when $m \notin \mathbb{Z} \bigwedge n \notin \mathbb{Z} \bigwedge p \notin \mathbb{Z} \bigwedge \frac{b}{b c-a d} > 0$

1: $\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx$ when $m \notin \mathbb{Z} \bigwedge n \notin \mathbb{Z} \bigwedge p \notin \mathbb{Z} \bigwedge \frac{b}{b c-a d} > 0 \bigwedge \frac{b}{b e-a f} > 0$

Rule 1.1.1.3.A.3.1.1: If $m \notin \mathbb{Z} \bigwedge n \notin \mathbb{Z} \bigwedge p \notin \mathbb{Z} \bigwedge \frac{b}{b c-a d} > 0 \bigwedge \frac{b}{b e-a f} > 0$**, then**

$$\int (a+b x)^m (c+d x)^n (e+f x)^p \, dx \longrightarrow$$

$$\frac{(a+b x)^{m+1}}{b (m+1) \left(\frac{b}{b c-a d}\right)^n \left(\frac{b}{b e-a f}\right)^p} \text{AppellF1}\left[m+1, -n, -p, m+2, -\frac{d (a+b x)}{b c-a d}, -\frac{f (a+b x)}{b e-a f}\right]$$

Program code:

```
Int[(a_+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_,x_Symbol] :=
  (a+b*x)^(m+1)/(b*(m+1)*(b/(b*c-a*d))^n*(b/(b*e-a*f))^p)*AppellF1[m+1,-n,-p,m+2,-d*(a+b*x)/(b*c-a*d),-f*(a+b*x)/(b*e-a*f)] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && Not[IntegerQ[p]] &&
  GtQ[b/(b*c-a*d),0] && GtQ[b/(b*e-a*f),0] &&
  Not[GtQ[d/(d*a-c*b),0] && GtQ[d/(d*e-c*f),0] && SimplerQ[c+d*x,a+b*x]] &&
  Not[GtQ[f/(f*a-e*b),0] && GtQ[f/(f*c-e*d),0] && SimplerQ[e+f*x,a+b*x]]
```

2: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m\notin\mathbb{Z}\bigwedge n\notin\mathbb{Z}\bigwedge p\notin\mathbb{Z}\bigwedge \frac{b}{b\,c-a\,d}>0\bigwedge \frac{b}{b\,e-a\,f}\ngtr 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(e+f\,x)^p}{\left(\frac{b\,(e+f\,x)}{b\,e-a\,f}\right)^p}==0$
- **Rule 1.1.1.3.A.3.1.2: If** $m\notin\mathbb{Z}\bigwedge n\notin\mathbb{Z}\bigwedge p\notin\mathbb{Z}\bigwedge \frac{b}{b\,c-a\,d}>0\bigwedge \frac{b}{b\,e-a\,f}\ngtr 0$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\;\longrightarrow\;\frac{(e+f\,x)^{\text{FracPart}[p]}}{\left(\frac{b}{b\,e-a\,f}\right)^{\text{IntPart}[p]}\left(\frac{b\,(e+f\,x)}{b\,e-a\,f}\right)^{\text{FracPart}[p]}}\int (a+b\,x)^m\,(c+d\,x)^n\left(\frac{b\,e}{b\,e-a\,f}+\frac{b\,f\,x}{b\,e-a\,f}\right)^p dx$$

- **Program code:**

```
Int[(a_+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_,x_Symbol] :=
  (e+f*x)^FracPart[p]/((b/(b*e-a*f))^IntPart[p]*(b*(e+f*x)/(b*e-a*f))^FracPart[p])*
    Int[(a+b*x)^m*(c+d*x)^n*(b*e/(b*e-a*f)+b*f*x/(b*e-a*f))^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && Not[IntegerQ[p]] &&
  GtQ[b/(b*c-a*d),0] && Not[GtQ[b/(b*e-a*f),0]]
```

2: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m\notin\mathbb{Z}\bigwedge n\notin\mathbb{Z}\bigwedge p\notin\mathbb{Z}\bigwedge \frac{b}{b\,c-a\,d}\ngtr 0$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \frac{(c+d\,x)^n}{\left(\frac{b\,(c+d\,x)}{b\,c-a\,d}\right)^n}==0$
- **Rule 1.1.1.3.A.3.2: If** $m\notin\mathbb{Z}\bigwedge n\notin\mathbb{Z}\bigwedge p\notin\mathbb{Z}\bigwedge \frac{b}{b\,c-a\,d}\ngtr 0$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\;\longrightarrow\;\frac{(c+d\,x)^{\text{FracPart}[n]}}{\left(\frac{b}{b\,c-a\,d}\right)^{\text{IntPart}[n]}\left(\frac{b\,(c+d\,x)}{b\,c-a\,d}\right)^{\text{FracPart}[n]}}\int (a+b\,x)^m\left(\frac{b\,c}{b\,c-a\,d}+\frac{b\,d\,x}{b\,c-a\,d}\right)^n(e+f\,x)^p\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_,x_Symbol] :=
  (c+d*x)^FracPart[n]/((b/(b*c-a*d))^IntPart[n]*(b*(c+d*x)/(b*c-a*d))^FracPart[n])*
    Int[(a+b*x)^m*(b*c/(b*c-a*d)+b*d*x/(b*c-a*d))^n*(e+f*x)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && Not[IntegerQ[p]] && Not[GtQ[b/(b*c-a*d),0]] &&
  Not[SimplerQ[c+d*x,a+b*x]] && Not[SimplerQ[e+f*x,a+b*x]]
```

S: $\int (a+b\,u)^m\,(c+d\,u)^n\,(e+f\,u)^p\,dx$ when $u == g+h\,x$

- **Derivation: Integration by substitution**
- **Rule 1.1.1.3.S: If $u == g+h\,x$, then**

$$\int (a+b\,u)^m\,(c+d\,u)^n\,(e+f\,u)^p\,dx \longrightarrow \frac{1}{h}\,\text{Subst}\left[\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx,\ x,\ u\right]$$

- **Program code:**

```
Int[(a_.+b_.*u_)^m_.*(c_.+d_.*u_)^n_.*(e_+f_.*u_)^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*x)^m*(c+d*x)^n*(e+f*x)^p,x],x,u] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && LinearQ[u,x] && NeQ[u,x]
```