Rules for integrands of the form $(a + b \operatorname{Tanh}[c + d x^n])^p$

1: $\int (a + b \operatorname{Tanh}[c + d x^n])^p \, dx$ when $\frac{1}{n} \in \mathbb{Z}^+ \bigwedge p \in \mathbb{Z}$

- **Derivation: Integration by substitution**
- **Basis: If** $-1 \leq n \leq 1 \land n \neq 0$**, then** $F[x^n] = \frac{1}{n} \operatorname{Subst}\left[x^{\frac{1}{n}-1} F[x], x, x^n\right] \partial_x x^n$
- **Note: If** $\frac{1}{n} \in \mathbb{Z}^-$**, resulting integrand is not integrable.**
- **Rule: If** $\frac{1}{n} \in \mathbb{Z}^+ \bigwedge p \in \mathbb{Z}$**, then**

$$\int (a + b \operatorname{Tanh}[c + d x^n])^p \, dx \longrightarrow \frac{1}{n} \operatorname{Subst}\left[\int x^{\frac{1}{n}-1} (a + b \operatorname{Tanh}[c + d x])^p \, dx, x, x^n\right]$$

- **Program code:**

```
Int[(a_.+b_.*Tanh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(1/n-1)*(a+b*Tanh[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,p},x] && IGtQ[1/n,0] && IntegerQ[p]
```

```
Int[(a_.+b_.*Coth[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(1/n-1)*(a+b*Coth[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,p},x] && IGtQ[1/n,0] && IntegerQ[p]
```

X: $\int (a + b \operatorname{Tanh}[c + d x^n])^p \, dx$

- **Rule:**

$$\int (a + b \operatorname{Tanh}[c + d x^n])^p \, dx \longrightarrow \int (a + b \operatorname{Tanh}[c + d x^n])^p \, dx$$

- **Program code:**

```
Int[(a_.+b_.*Tanh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Integral[(a+b*Tanh[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,n,p},x]
```

```
Int[(a_.+b_.*Coth[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Integral[(a+b*Coth[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,n,p},x]
```

S: $\int (a + b \operatorname{Tanh}[c + d u^n])^p \, dx$ when $u == e + f x$

Derivation: Integration by substitution

Rule: If $u == e + f x$, then

$$\int (a + b \operatorname{Tanh}[c + d u^n])^p \, dx \longrightarrow \frac{1}{f} \operatorname{Subst}\left[\int (a + b \operatorname{Tanh}[c + d x^n])^p \, dx,\ x,\ u\right]$$

Program code:

```
Int[(a_.+b_.*Tanh[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*Tanh[c+d*x^n])^p,x],x,u] /;
FreeQ[{a,b,c,d,n,p},x] && LinearQ[u,x] && NeQ[u,x]
```

```
Int[(a_.+b_.*Coth[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*Coth[c+d*x^n])^p,x],x,u] /;
FreeQ[{a,b,c,d,n,p},x] && LinearQ[u,x] && NeQ[u,x]
```

N: $\int (a + b \operatorname{Tanh}[u])^p \, dx$ when $u == c + d x^n$

Derivation: Algebraic normalization

Rule: If $u == c + d x^n$, then

$$\int (a + b \operatorname{Tanh}[u])^p \, dx \longrightarrow \int (a + b \operatorname{Tanh}[c + d x^n])^p \, dx$$

Program code:

```
Int[(a_.+b_.*Tanh[u_])^p_.,x_Symbol] :=
  Int[(a+b*Tanh[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

```
Int[(a_.+b_.*Coth[u_])^p_.,x_Symbol] :=
  Int[(a+b*Coth[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

Rules for integrands of the form $(e\,x)^m\,(a+b\,\text{Tanh}[c+d\,x^n])^p$

1. $\int x^m\,(a+b\,\text{Tanh}[c+d\,x^n])^p\,dx$

1: $\int x^m\,(a+b\,\text{Tanh}[c+d\,x^n])^p\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}^+\bigwedge p\in\mathbb{Z}$

Derivation: Integration by substitution

- **Basis: If** $\frac{m+1}{n}\in\mathbb{Z}$, **then** $x^m\,F[x^n]=\frac{1}{n}\,\text{Subst}\left[x^{\frac{m+1}{n}-1}\,F[x],\,x,\,x^n\right]\,\partial_x x^n$
- **Note: If** $\frac{m+1}{n}\in\mathbb{Z}^-$, **resulting integrand is not integrable.**
- **Rule: If** $\frac{m+1}{n}\in\mathbb{Z}^+\bigwedge p\in\mathbb{Z}$, **then**

$$\int x^m\,(a+b\,\text{Tanh}[c+d\,x^n])^p\,dx\;\longrightarrow\;\frac{1}{n}\,\text{Subst}\left[\int x^{\frac{m+1}{n}-1}\,(a+b\,\text{Tanh}[c+d\,x])^p\,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Tanh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*Tanh[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,m,n,p},x] && IGtQ[Simplify[(m+1)/n],0] && IntegerQ[p]
```

```
Int[x_^m_.*(a_.+b_.*Coth[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*Coth[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,m,n,p},x] && IGtQ[Simplify[(m+1)/n],0] && IntegerQ[p]
```

2: $\int x^m\,\text{Tanh}[c+d\,x^n]^2\,dx$

- **Note: Although this rule reduces the degree of the tangent factor, the resulting integral is not integrable unless** $\frac{m+1}{n}\in\mathbb{Z}^+$.

Rule:

$$\int x^m\,\text{Tanh}[c+d\,x^n]^2\,dx\;\longrightarrow\;-\frac{x^{m-n+1}\,\text{Tanh}[c+d\,x^n]}{d\,n}+\int x^m\,dx-\frac{m-n+1}{d\,n}\int x^{m-n}\,\text{Tanh}[c+d\,x^n]\,dx$$

Program code:

```
Int[x_^m_.*Tanh[c_.+d_.*x_^n_]^2,x_Symbol] :=
  -x^(m-n+1)*Tanh[c+d*x^n]/(d*n) + Int[x^m,x] + (m-n+1)/(d*n)*Int[x^(m-n)*Tanh[c+d*x^n],x] /;
FreeQ[{c,d,m,n},x]
```

```
Int[x_^m_.*Coth[c_.+d_.*x_^n_]^2,x_Symbol] :=
  -x^(m-n+1)*Coth[c+d*x^n]/(d*n) + Int[x^m,x] + (m-n+1)/(d*n)*Int[x^(m-n)*Coth[c+d*x^n],x] /;
FreeQ[{c,d,m,n},x]
```

X: $\int x^m \, (a + b \operatorname{Tanh}[c + d x^n])^p \, dx$

Rule:

$$\int x^m \, (a + b \operatorname{Tanh}[c + d x^n])^p \, dx \longrightarrow \int x^m \, (a + b \operatorname{Tanh}[c + d x^n])^p \, dx$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Tanh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Integral[x^m*(a+b*Tanh[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x]
```

```
Int[x_^m_.*(a_.+b_.*Coth[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Integral[x^m*(a+b*Coth[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x]
```

2: $\int (e\, x)^m \, (a + b \operatorname{Tanh}[c + d x^n])^p \, dx$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(e\, x)^m}{x^m} == 0$

Rule:

$$\int (e\, x)^m \, (a + b \operatorname{Tanh}[c + d x^n])^p \, dx \longrightarrow \frac{e^{\operatorname{IntPart}[m]} \, (e\, x)^{\operatorname{FracPart}[m]}}{x^{\operatorname{FracPart}[m]}} \int x^m \, (a + b \operatorname{Tanh}[c + d x^n])^p \, dx$$

Program code:

```
Int[(e_*x_)^m_.*(a_.+b_.*Tanh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Tanh[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```

```
Int[(e_*x_)^m_.*(a_.+b_.*Coth[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Coth[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```

N: $\int (e\,x)^m\,(a+b\,\text{Tanh}[u])^p\,dx$ when $u == c+d\,x^n$

Derivation: Algebraic normalization

- **Rule: If** $u == c+d\,x^n$**, then**

$$\int (e\,x)^m\,(a+b\,\text{Tanh}[u])^p\,dx \longrightarrow \int (e\,x)^m\,(a+b\,\text{Tanh}[c+d\,x^n])^p\,dx$$

- **Program code:**

```
Int[(e_*x_)^m_.*(a_.+b_.*Tanh[u_])^p_.,x_Symbol] :=
  Int[(e*x)^m*(a+b*Tanh[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,e,m,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

```
Int[(e_*x_)^m_.*(a_.+b_.*Coth[u_])^p_.,x_Symbol] :=
  Int[(e*x)^m*(a+b*Coth[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,e,m,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

Rules for integrands of the form $x^m \operatorname{Sech}[a + b x^n]^p \operatorname{Tanh}[a + b x^n]$

1: $\int x^m \operatorname{Sech}[a + b x^n]^p \operatorname{Tanh}[a + b x^n]\, dx$ when $n \in \mathbb{Z} \land m - n \geq 0$

- **Derivation: Integration by parts**
- **Note: Dummy exponent $q = 1$ required in program code so InputForm of integrand is recognized.**
- **Rule: If $n \in \mathbb{Z} \land m - n \geq 0$, then**

$$\int x^m \operatorname{Sech}[a + b x^n]^p \operatorname{Tanh}[a + b x^n]\, dx \longrightarrow -\frac{x^{m-n+1} \operatorname{Sech}[a + b x^n]^p}{b n p} + \frac{m - n + 1}{b n p} \int x^{m-n} \operatorname{Sech}[a + b x^n]^p\, dx$$

- **Program code:**

```
Int[x_^m_.*Sech[a_.+b_.*x_^n_.]^p_.*Tanh[a_.+b_.*x_^n_.]^q_.,x_Symbol] :=
  -x^(m-n+1)*Sech[a+b*x^n]^p/(b*n*p) +
  (m-n+1)/(b*n*p)*Int[x^(m-n)*Sech[a+b*x^n]^p,x] /;
FreeQ[{a,b,p},x] && RationalQ[m] && IntegerQ[n] && GeQ[m-n,0] && EqQ[q,1]
```

```
Int[x_^m_.*Csch[a_.+b_.*x_^n_.]^p_.*Coth[a_.+b_.*x_^n_.]^q_.,x_Symbol] :=
  -x^(m-n+1)*Csch[a+b*x^n]^p/(b*n*p) +
  (m-n+1)/(b*n*p)*Int[x^(m-n)*Csch[a+b*x^n]^p,x] /;
FreeQ[{a,b,p},x] && RationalQ[m] && IntegerQ[n] && GeQ[m-n,0] && EqQ[q,1]
```