Rules for integrands of the form $(d\,x)^m\,P_q[x]\,\left(a+b\,x^2+c\,x^4\right)^p$

1: $\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^2+c\,x^4\right)^p\,dx$ when $\neg\,P_q\left[x^2\right]$

- Derivation: Algebraic expansion
- Basis: $P_q[x] = \sum_{k=0}^{q/2+1} P_q[x,\ 2\,k]\ x^{2\,k} + x\,\sum_{k=0}^{(q-1)/2+1} P_q[x,\ 2\,k+1]\ x^{2\,k}$
- Note: This rule transforms $P_q[x]$ into a sum of the form $Q_r\left[x^2\right] + x\,R_s\left[x^2\right]$.
- Rule 1.2.2.6.3: If $\neg\,P_q\left[x^2\right]$, then

$$\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^2+c\,x^4\right)^p\,dx\ \longrightarrow\ \int (d\,x)^m\left(\sum_{k=0}^{\frac{q}{2}+1} P_q[x,\ 2\,k]\ x^{2\,k}\right)\left(a+b\,x^2+c\,x^4\right)^p\,dx+\frac{1}{d}\int (d\,x)^{m+1}\left(\sum_{k=0}^{\frac{q-1}{2}+1} P_q[x,\ 2\,k+1]\ x^{2\,k}\right)\left(a+b\,x^2+c\,x^4\right)^p\,dx$$

- Program code:

```
Int[(d_.*x_)^m_.*Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  Module[{q=Expon[Pq,x],k},
  Int[(d*x)^m*Sum[Coeff[Pq,x,2*k]*x^(2*k),{k,0,q/2+1}]*(a+b*x^2+c*x^4)^p,x] +
  1/d*Int[(d*x)^(m+1)*Sum[Coeff[Pq,x,2*k+1]*x^(2*k),{k,0,(q-1)/2+1}]*(a+b*x^2+c*x^4)^p,x]] /;
FreeQ[{a,b,c,d,m,p},x] && PolyQ[Pq,x] && Not[PolyQ[Pq,x^2]]
```

2: $\int x^m\,P_q\left[x^2\right]\,\left(a+b\,x^2+c\,x^4\right)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}$

- Derivation: Integration by substitution
- Basis: If $\frac{m-1}{2}\in\mathbb{Z}$, then $x^m\,F\left[x^2\right] = \frac{1}{2}\,\text{Subst}\left[x^{\frac{m-1}{2}}\,F[x],\ x,\ x^2\right]\,\partial_x x^2$
- Rule 1.2.2.6.4: If $\frac{m-1}{2}\in\mathbb{Z}$, then

$$\int x^m\,P_q\left[x^2\right]\,\left(a+b\,x^2+c\,x^4\right)^p\,dx\ \longrightarrow\ \frac{1}{2}\,\text{Subst}\left[\int x^{\frac{m-1}{2}}\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx,\ x,\ x^2\right]$$

- Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  1/2*Subst[Int[x^((m-1)/2)*SubstFor[x^2,Pq,x]*(a+b*x+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,b,c,p},x] && PolyQ[Pq,x^2] && IntegerQ[(m-1)/2]
```

3: $\int (d\,x)^m\,P_q\left[x^2\right]\,\left(a+b\,x^2+c\,x^4\right)^p\,dx$ when $p+2\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.2.6.1: If $p+2\in\mathbb{Z}^+$, then

$$\int (d\,x)^m\,P_q\left[x^2\right]\,\left(a+b\,x^2+c\,x^4\right)^p\,dx\;\longrightarrow\;\int \text{ExpandIntegrand}\left[(d\,x)^m\,P_q\left[x^2\right]\,\left(a+b\,x^2+c\,x^4\right)^p,\,x\right]dx$$

Program code:

```
Int[(d_.*x_)^m_.*Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d*x)^m*Pq*(a+b*x^2+c*x^4)^p,x],x] /;
FreeQ[{a,b,c,d,m},x] && PolyQ[Pq,x^2] && IGtQ[p,-2]
```

4: $\int (d\,x)^m\,P_q\left[x^2\right]\,\left(a+b\,x^2+c\,x^4\right)^p\,dx$ when $P_q[x,\,0]==0$

Derivation: Algebraic expansion

Rule 1.2.2.6.2: If $P_q[x,\,0]==0$, then

$$\int (d\,x)^m\,P_q\left[x^2\right]\,\left(a+b\,x^2+c\,x^4\right)^p\,dx\;\longrightarrow\;\frac{1}{d^2}\int (d\,x)^{m+2}\,\frac{P_q\left[x^2\right]}{x^2}\,\left(a+b\,x^2+c\,x^4\right)^p\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  1/d^2*Int[(d*x)^(m+2)*ExpandToSum[Pq/x^2,x]*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,d,m,p},x] && PolyQ[Pq,x^2] && EqQ[Coeff[Pq,x,0],0]
```

5: $\int (d\,x)^m \left(e+f\,x^2+g\,x^4\right) \left(a+b\,x^2+c\,x^4\right)^p dx$ when $a\,f\,(m+1)-b\,e\,(m+2\,p+3)=0\,\land\,a\,g\,(m+1)-c\,e\,(m+4\,p+5)=0\,\land\,m\neq-1$

Rule 1.2.2.6.5: If $a\,f\,(m+1)-b\,e\,(m+2\,p+3)=0\,\land\,a\,g\,(m+1)-c\,e\,(m+4\,p+5)=0\,\land\,m\neq-1$**, then**

$$\int (d\,x)^m \left(e+f\,x^2+g\,x^4\right) \left(a+b\,x^2+c\,x^4\right)^p dx \to \frac{e\,(d\,x)^{m+1}\left(a+b\,x^2+c\,x^4\right)^{p+1}}{a\,d\,(m+1)}$$

Program code:

```
Int[(d_.*x_)^m_.*Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  With[{e=Coeff[Pq,x,0],f=Coeff[Pq,x,2],g=Coeff[Pq,x,4]},
  e*(d*x)^(m+1)*(a+b*x^2+c*x^4)^(p+1)/(a*d*(m+1)) /;
 EqQ[a*f*(m+1)-b*e*(m+2*p+3),0] && EqQ[a*g*(m+1)-c*e*(m+4*p+5),0] && NeQ[m,-1]] /;
FreeQ[{a,b,c,d,m,p},x] && PolyQ[Pq,x^2] && EqQ[Expon[Pq,x],4]
```

6: $\int (d\,x)^m\,P_q\left[x^2\right] \left(a+b\,x^2+c\,x^4\right)^p dx$ when $q>1\,\land\,b^2-4\,a\,c=0$

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c=0$**, then** $\partial_x \frac{(a+b\,x^2+c\,x^4)^p}{(b+2\,c\,x^2)^{2\,p}}=0$

Rule 1.2.2.6.7: If $q>1\,\land\,b^2-4\,a\,c=0$**, then**

$$\int (d\,x)^m\,P_q\left[x^2\right] \left(a+b\,x^2+c\,x^4\right)^p dx \to \frac{\left(a+b\,x^2+c\,x^4\right)^{\text{FracPart}[p]}}{(4\,c)^{\text{IntPart}[p]}\left(b+2\,c\,x^2\right)^{2\,\text{FracPart}[p]}} \int (d\,x)^m\,P_q\left[x^2\right]\left(b+2\,c\,x^2\right)^{2\,p} dx$$

Program code:

```
Int[(d_.*x_)^m_.*Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (a+b*x^2+c*x^4)^FracPart[p]/((4*c)^IntPart[p]*(b+2*c*x^2)^(2*FracPart[p]))*Int[(d*x)^m*Pq*(b+2*c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,d,m,p},x] && PolyQ[Pq,x^2] && GtQ[Expon[Pq,x^2],1] && EqQ[b^2-4*a*c,0]
```

7. $\int x^m P_q[x^2] \left(a+b x^2+c x^4\right)^p dx$ when $q>1 \bigwedge b^2-4 a c \neq 0 \bigwedge p<-1 \bigwedge \frac{m}{2} \in \mathbb{Z}$

1: $\int x^m P_q[x^2] \left(a+b x^2+c x^4\right)^p dx$ when $q>1 \bigwedge b^2-4 a c \neq 0 \bigwedge p<-1 \bigwedge \frac{m}{2} \in \mathbb{Z}^+$

Derivation: Algebraic expansion and trinomial recurrence 2b

Rule 1.2.2.6.8.1: If $q>1 \bigwedge b^2-4 a c \neq 0 \bigwedge p<-1 \bigwedge \frac{m}{2} \in \mathbb{Z}^+$**, let** $Q \to \text{PolynomialQuotient}\left[x^m P_q[x^2],\ a+b x^2+c x^4,\ x\right]$ **and** $d+e x^2 \to \text{PolynomialRemainder}\left[x^m P_q[x^2],\ a+b x^2+c x^4,\ x\right]$**, then**

$$\int x^m P_q[x^2] \left(a+b x^2+c x^4\right)^p dx \longrightarrow$$

$$\int \left(d+e x^2\right) \left(a+b x^2+c x^4\right)^p dx + \int Q \left(a+b x^2+c x^4\right)^{p+1} dx \longrightarrow$$

$$\frac{x \left(a+b x^2+c x^4\right)^{p+1} \left(a b e-d \left(b^2-2 a c\right)-c (b d-2 a e) x^2\right)}{2 a (p+1) \left(b^2-4 a c\right)} + \frac{1}{2 a (p+1) \left(b^2-4 a c\right)} \int \left(a+b x^2+c x^4\right)^{p+1} \cdot \left(2 a (p+1) \left(b^2-4 a c\right) Q + b^2 d (2 p+3) - 2 a c d (4 p+5) - a b e + c (4 p+7) (b d-2 a e) x^2\right) dx$$

Program code:

```
Int[x_^m_*Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  With[{d=Coeff[PolynomialRemainder[x^m*Pq,a+b*x^2+c*x^4,x],x,0],
        e=Coeff[PolynomialRemainder[x^m*Pq,a+b*x^2+c*x^4,x],x,2]},
  x*(a+b*x^2+c*x^4)^(p+1)*(a*b*e-d*(b^2-2*a*c)-c*(b*d-2*a*e)*x^2)/(2*a*(p+1)*(b^2-4*a*c)) +
  1/(2*a*(p+1)*(b^2-4*a*c))*Int[(a+b*x^2+c*x^4)^(p+1)*
    ExpandToSum[2*a*(p+1)*(b^2-4*a*c)*PolynomialQuotient[x^m*Pq,a+b*x^2+c*x^4,x]+
      b^2*d*(2*p+3)-2*a*c*d*(4*p+5)-a*b*e+c*(4*p+7)*(b*d-2*a*e)*x^2,x],x]] /;
FreeQ[{a,b,c},x] && PolyQ[Pq,x^2] && GtQ[Expon[Pq,x^2],1] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && IGtQ[m/2,0]
```

2: $\int x^m P_q[x^2] \left(a+b x^2+c x^4\right)^p dx$ when $q>1 \bigwedge b^2-4ac\neq 0 \bigwedge p<-1 \bigwedge \frac{m}{2}\in\mathbb{Z}^-$

Derivation: Algebraic expansion and trinomial recurrence 2b

Rule 1.2.2.6.8.2: If $q>1 \bigwedge b^2-4ac\neq 0 \bigwedge p<-1 \bigwedge \frac{m}{2}\in\mathbb{Z}^-$, **let** $Q\to \text{PolynomialQuotient}\left[x^m P_q[x^2],\ a+b x^2+c x^4,\ x\right]$ **and** $d+e x^2\to \text{PolynomialRemainder}\left[x^m P_q[x^2],\ a+b x^2+c x^4,\ x\right]$, **then**

$$\int x^m P_q[x^2] \left(a+b x^2+c x^4\right)^p dx \longrightarrow$$

$$\int \left(d+e x^2\right) \left(a+b x^2+c x^4\right)^p dx+\int Q \left(a+b x^2+c x^4\right)^{p+1} dx \longrightarrow$$

$$\frac{x \left(a+b x^2+c x^4\right)^{p+1} \left(a b e-d \left(b^2-2 a c\right)-c\,(b d-2 a e)\,x^2\right)}{2 a\,(p+1) \left(b^2-4 a c\right)}+$$
$$\frac{1}{2 a\,(p+1) \left(b^2-4 a c\right)} \int x^m \left(a+b x^2+c x^4\right)^{p+1} \cdot$$
$$\left(2 a\,(p+1) \left(b^2-4 a c\right) x^{-m} Q+\left(b^2 d\,(2 p+3)-2 a c d\,(4 p+5)-a b e\right) x^{-m}+c\,(4 p+7)\,(b d-2 a e)\,x^{2-m}\right) dx$$

Program code:

```
Int[x_^m_*Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  With[{d=Coeff[PolynomialRemainder[x^m*Pq,a+b*x^2+c*x^4,x],x,0],
        e=Coeff[PolynomialRemainder[x^m*Pq,a+b*x^2+c*x^4,x],x,2]},
  x*(a+b*x^2+c*x^4)^(p+1)*(a*b*e-d*(b^2-2*a*c)-c*(b*d-2*a*e)*x^2)/(2*a*(p+1)*(b^2-4*a*c)) +
  1/(2*a*(p+1)*(b^2-4*a*c))*Int[x^m*(a+b*x^2+c*x^4)^(p+1)*
    ExpandToSum[2*a*(p+1)*(b^2-4*a*c)*x^(-m)*PolynomialQuotient[x^m*Pq,a+b*x^2+c*x^4,x]+
      (b^2*d*(2*p+3)-2*a*c*d*(4*p+5)-a*b*e)*x^(-m)+c*(4*p+7)*(b*d-2*a*e)*x^(2-m),x],x]] /;
FreeQ[{a,b,c},x] && PolyQ[Pq,x^2] && GtQ[Expon[Pq,x^2],1] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && ILtQ[m/2,0]
```

x: $\int x^m P_q[x^2] \left(a+b x^2+c x^4\right)^p dx$ when $q>1 \bigwedge b^2-4ac\neq 0 \bigwedge p<-1 \bigwedge \frac{m-1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion and trinomial recurrence 2b

Note: Better to use the substitution $x\to x^2$.

Rule 1.2.2.6.8.2: If $q>1 \bigwedge b^2-4ac\neq 0 \bigwedge p<-1 \bigwedge \frac{m-1}{2}\in\mathbb{Z}$, **let** $Q\to \text{PolynomialQuotient}\left[x^m P_q[x^2],\ a+b x^2+c x^4,\ x\right]$ **and** $d x+e x^3\to \text{PolynomialRemainder}\left[x^m P_q[x^2],\ a+b x^2+c x^4,\ x\right]$, **then**

$$\int x^m P_q\left[x^2\right] \left(a + b x^2 + c x^4\right)^p dx \longrightarrow$$

$$\int \left(d x + e x^3\right) \left(a + b x^2 + c x^4\right)^p dx + \int Q \left(a + b x^2 + c x^4\right)^{p+1} dx \longrightarrow$$

$$\frac{x^2 \left(a + b x^2 + c x^4\right)^{p+1} \left(a b e - d \left(b^2 - 2 a c\right) - c (b d - 2 a e) x^2\right)}{2 a (p + 1) \left(b^2 - 4 a c\right)} +$$

$$\frac{1}{a (p + 1) \left(b^2 - 4 a c\right)} \int x^m \left(a + b x^2 + c x^4\right)^{p+1} \cdot$$

$$\left(a (p + 1) \left(b^2 - 4 a c\right) x^{-m} Q + \left(b^2 d (p + 2) - 2 a c d (2 p + 3) - a b e\right) x^{1-m} + 2 c (p + 2) (b d - 2 a e) x^{3-m}\right) dx$$

- **Program code:**

```
(* Int[x_^m_.*Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  With[{d=Coeff[PolynomialRemainder[x^m*Pq,a+b*x^2+c*x^4,x],x,1],
        e=Coeff[PolynomialRemainder[x^m*Pq,a+b*x^2+c*x^4,x],x,3]},
  x^2*(a+b*x^2+c*x^4)^(p+1)*(a*b*e-d*(b^2-2*a*c)-c*(b*d-2*a*e)*x^2)/(2*a*(p+1)*(b^2-4*a*c)) +
  1/(a*(p+1)*(b^2-4*a*c))*Int[x^m*(a+b*x^2+c*x^4)^(p+1)*
    ExpandToSum[a*(p+1)*(b^2-4*a*c)*x^(-m)*PolynomialQuotient[x^m*Pq,a+b*x^2+c*x^4,x]+
      (b^2*d*(p+2)-2*a*c*d*(2*p+3)-a*b*e)*x^(1-m)+2*c*(p+2)*(b*d-2*a*e)*x^(3-m),x],x]] /;
FreeQ[{a,b,c},x] && PolyQ[Pq,x^2] && GtQ[Expon[Pq,x^2],1] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && IntegerQ[(m-1)/2] *)
```

U: $\int (d x)^m P_q[x] \left(a + b x^2 + c x^4\right)^p dx$

- **Rule 1.2.2.6.U:**

$$\int (d x)^m P_q[x] \left(a + b x^2 + c x^4\right)^p dx \longrightarrow \int (d x)^m P_q[x] \left(a + b x^2 + c x^4\right)^p dx$$

- **Program code:**

```
Int[(d_.*x_)^m_.*Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Unintegrable[(d*x)^m*Pq*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,d,m,p},x] && PolyQ[Pq,x]
```