Rules for integrands involving polylogarithms

1. $\int u\,\text{PolyLog}[n,\,a\,(b\,x^p)^q]\,dx$

   1. $\int \text{PolyLog}[n,\,a\,(b\,x^p)^q]\,dx$

      1. $\int \text{PolyLog}[n,\,a\,(b\,x^p)^q]\,dx$ when $n>0$

         x: $\int \text{PolyLog}[2,\,a\,(b\,x^p)^q]\,dx$

- **Derivation: Integration by parts**
- **Note: This rule not necessary for host systems, like *Mathematica*, that automatically simplify** $\text{PolyLog}[1,\,z]$ **to** $-\text{Log}[1-z]$**.**
- **Rule:**

$$\int \text{PolyLog}[2,\,a\,(b\,x^p)^q]\,dx \longrightarrow x\,\text{PolyLog}[2,\,a\,(b\,x^p)^q]+p\,q\int \text{Log}[1-a\,(b\,x^p)^q]\,dx$$

- **Program code:**

```
(* Int[PolyLog[2,a_.*(b_.*x_^p_.)^q_.],x_Symbol] :=
  x*PolyLog[2,a*(b*x^p)^q] + p*q*Int[Log[1-a*(b*x^p)^q],x] /;
FreeQ[{a,b,p,q},x] *)
```

2: $\int \text{PolyLog}[n,\,a\,(b\,x^p)^q]\,dx$ when $n>0$

- **Derivation: Integration by parts**
- **Rule: If** $n>0$**, then**

$$\int \text{PolyLog}[n,\,a\,(b\,x^p)^q]\,dx \longrightarrow x\,\text{PolyLog}[n,\,a\,(b\,x^p)^q]-p\,q\int \text{PolyLog}[n-1,\,a\,(b\,x^p)^q]\,dx$$

- **Program code:**

```
Int[PolyLog[n_,a_.*(b_.*x_^p_.)^q_.],x_Symbol] :=
  x*PolyLog[n,a*(b*x^p)^q] - p*q*Int[PolyLog[n-1,a*(b*x^p)^q],x] /;
FreeQ[{a,b,p,q},x] && GtQ[n,0]
```

2: $\int \text{PolyLog}[n, a (b x^p)^q] \, dx$ when $n < -1$

Derivation: Inverted integration by parts

Rule: If $n < -1$, then

$$\int \text{PolyLog}[n, a (b x^p)^q] \, dx \to \frac{x \, \text{PolyLog}[n+1, a (b x^p)^q]}{p q} - \frac{1}{p q} \int \text{PolyLog}[n+1, a (b x^p)^q] \, dx$$

Program code:

```
Int[PolyLog[n_,a_.*(b_.*x_^p_.)^q_.],x_Symbol] :=
  x*PolyLog[n+1,a*(b*x^p)^q]/(p*q) - 1/(p*q)*Int[PolyLog[n+1,a*(b*x^p)^q],x] /;
FreeQ[{a,b,p,q},x] && LtQ[n,-1]
```

U: $\int \text{PolyLog}[n, a (b x^p)^q] \, dx$

Rule:

$$\int \text{PolyLog}[n, a (b x^p)^q] \, dx \to \int \text{PolyLog}[n, a (b x^p)^q] \, dx$$

Program code:

```
Int[PolyLog[n_,a_.*(b_.*x_^p_.)^q_.],x_Symbol] :=
  Unintegrable[PolyLog[n,a*(b*x^p)^q],x] /;
FreeQ[{a,b,n,p,q},x]
```

2. $\int (d\,x)^m \,\text{PolyLog}[n,\ a\,(b\,x^p)^q]\,dx$

1: $\int \frac{\text{PolyLog}[n,\ a\,(b\,x^p)^q]}{x}\,dx$

Derivation: Primitive rule

Basis: $\frac{\partial \text{Li}_n(z)}{\partial z} = \frac{\text{Li}_{n-1}(z)}{z}$

Rule:

$$\int \frac{\text{PolyLog}[n,\ a\,(b\,x^p)^q]}{x}\,dx \longrightarrow \frac{\text{PolyLog}[n+1,\ a\,(b\,x^p)^q]}{p\,q}$$

Program code:

```
Int[PolyLog[n_,c_.*(a_.+b_.*x_)^p_.]/(d_.+e_.*x_),x_Symbol] :=
  PolyLog[n+1,c*(a+b*x)^p]/(e*p) /;
FreeQ[{a,b,c,d,e,n,p},x] && EqQ[b*d,a*e]
```

```
Int[PolyLog[n_,a_.*(b_.*x_^p_.)^q_.]/x_,x_Symbol] :=
  PolyLog[n+1,a*(b*x^p)^q]/(p*q) /;
FreeQ[{a,b,n,p,q},x]
```

2. $\int (d\,x)^m \,\text{PolyLog}[n,\ a\,(b\,x^p)^q]\,dx$ when $m \neq -1$

1: $\int (d\,x)^m \,\text{PolyLog}[n,\ a\,(b\,x^p)^q]\,dx$ when $m \neq -1 \wedge n > 0$

Derivation: Integration by parts

Rule: If $m \neq -1 \wedge n > 0$**, then**

$$\int (d\,x)^m \,\text{PolyLog}[n,\ a\,(b\,x^p)^q]\,dx \longrightarrow \frac{(d\,x)^{m+1}\,\text{PolyLog}[n,\ a\,(b\,x^p)^q]}{d\,(m+1)} - \frac{p\,q}{m+1}\int (d\,x)^m \,\text{PolyLog}[n-1,\ a\,(b\,x^p)^q]\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*PolyLog[n_,a_.*(b_.*x_^p_.)^q_.],x_Symbol] :=
  (d*x)^(m+1)*PolyLog[n,a*(b*x^p)^q]/(d*(m+1)) -
  p*q/(m+1)*Int[(d*x)^m*PolyLog[n-1,a*(b*x^p)^q],x] /;
FreeQ[{a,b,d,m,p,q},x] && NeQ[m,-1] && GtQ[n,0]
```

2: $\int (d\,x)^m\,\text{PolyLog}[n,\,a\,(b\,x^p)^q]\,dx$ when $m \neq -1 \land n < -1$

Derivation: Inverted integration by parts

Rule: If $m \neq -1 \land n < -1$, then

$$\int (d\,x)^m\,\text{PolyLog}[n,\,a\,(b\,x^p)^q]\,dx \longrightarrow \frac{(d\,x)^{m+1}\,\text{PolyLog}[n+1,\,a\,(b\,x^p)^q]}{d\,p\,q} - \frac{m+1}{p\,q}\int (d\,x)^m\,\text{PolyLog}[n+1,\,a\,(b\,x^p)^q]\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*PolyLog[n_,a_.*(b_.*x_^p_.)^q_.],x_Symbol] :=
  (d*x)^(m+1)*PolyLog[n+1,a*(b*x^p)^q]/(d*p*q) -
  (m+1)/(p*q)*Int[(d*x)^m*PolyLog[n+1,a*(b*x^p)^q],x] /;
FreeQ[{a,b,d,m,p,q},x] && NeQ[m,-1] && LtQ[n,-1]
```

U: $\int (d\,x)^m\,\text{PolyLog}[n,\,a\,(b\,x^p)^q]\,dx$

Rule:

$$\int (d\,x)^m\,\text{PolyLog}[n,\,a\,(b\,x^p)^q]\,dx \longrightarrow \int (d\,x)^m\,\text{PolyLog}[n,\,a\,(b\,x^p)^q]\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*PolyLog[n_,a_.*(b_.*x_^p_.)^q_.],x_Symbol] :=
  Unintegrable[(d*x)^m*PolyLog[n,a*(b*x^p)^q],x] /;
FreeQ[{a,b,d,m,n,p,q},x]
```

3: $\int \frac{\text{Log}[c\,x^m]^r\,\text{PolyLog}[n,\,a\,(b\,x^p)^q]}{x}\,dx$ when $r>0$

Derivation: Integration by parts

Rule: If $r>0$, then

$$\int \frac{\text{Log}[c\,x^m]^r\,\text{PolyLog}[n,\,a\,(b\,x^p)^q]}{x}\,dx \longrightarrow$$
$$\frac{\text{Log}[c\,x^m]^r\,\text{PolyLog}[n+1,\,a\,(b\,x^p)^q]}{p\,q} - \frac{m\,r}{p\,q}\int \frac{\text{Log}[c\,x^m]^{r-1}\,\text{PolyLog}[n+1,\,a\,(b\,x^p)^q]}{x}\,dx$$

Program code:

```
Int[Log[c_.*x_^m_.]^r_.*PolyLog[n_,a_.*(b_.*x_^p_.)^q_.]/x_,x_Symbol] :=
  Log[c*x^m]^r*PolyLog[n+1,a*(b*x^p)^q]/(p*q) -
  m*r/(p*q)*Int[Log[c*x^m]^(r-1)*PolyLog[n+1,a*(b*x^p)^q]/x,x] /;
FreeQ[{a,b,c,m,n,q,r},x] && GtQ[r,0]
```

2. $\int u\,\text{PolyLog}[n,\ c\ (a+b\,x)^p]\,dx$

1: $\int \text{PolyLog}[n,\ c\ (a+b\,x)^p]\,dx$ when $n>0$

Derivation: Integration by parts and algebraic expansion

- **Basis:** $\partial_x\,\text{PolyLog}[n,\ c\ (a+b\,x)^p] == \frac{b\,p\,\text{PolyLog}[n-1,c\ (a+b\,x)^p]}{a+b\,x}$
- **Basis:** $\frac{x}{a+b\,x} == \frac{1}{b} - \frac{a}{b\ (a+b\,x)}$
- **Rule: If** $n>0$, **then**

$$\int \text{PolyLog}[n,\ c\ (a+b\,x)^p]\,dx \longrightarrow$$

$$x\,\text{PolyLog}[n,\ c\ (a+b\,x)^p] - b\,p\int\frac{x\,\text{PolyLog}[n-1,\ c\ (a+b\,x)^p]}{a+b\,x}\,dx \longrightarrow$$

$$x\,\text{PolyLog}[n,\ c\ (a+b\,x)^p] - p\int \text{PolyLog}[n-1,\ c\ (a+b\,x)^p]\,dx + a\,p\int\frac{\text{PolyLog}[n-1,\ c\ (a+b\,x)^p]}{a+b\,x}\,dx$$

- **Program code:**

```
Int[PolyLog[n_,c_.*(a_.+b_.*x_)^p_.],x_Symbol] :=
  x*PolyLog[n,c*(a+b*x)^p] -
  p*Int[PolyLog[n-1,c*(a+b*x)^p],x] +
  a*p*Int[PolyLog[n-1,c*(a+b*x)^p]/(a+b*x),x] /;
FreeQ[{a,b,c,p},x] && GtQ[n,0]
```

2. $\int (d+e\,x)^m\,\text{PolyLog}[n,\,c\,(a+b\,x)^p]\,dx$

 1. $\int (d+e\,x)^m\,\text{PolyLog}[2,\,c\,(a+b\,x)]\,dx$

  1. $\int \frac{\text{PolyLog}[2,\,c\,(a+b\,x)]}{d+e\,x}\,dx$

   1: $\int \frac{\text{PolyLog}[2,\,c\,(a+b\,x)]}{d+e\,x}\,dx$ when $c\,(b\,d-a\,e)+e=0$

Derivation: Integration by parts

- **Basis: If** $c\,(b\,d-a\,e)+e=0$, **then** $\frac{1}{d+e\,x}=\partial_x\frac{\text{Log}[1-a\,c-b\,c\,x]}{e}$
- **Basis:** $\partial_x\,\text{PolyLog}[2,\,c\,(a+b\,x)]=-\frac{b\,\text{Log}[1-a\,c-b\,c\,x]}{a+b\,x}$

– **Rule: If** $c\,(b\,d-a\,e)+e=0$, **then**

$$\int \frac{\text{PolyLog}[2,\,c\,(a+b\,x)]}{d+e\,x}\,dx\;\longrightarrow\;\frac{\text{Log}[1-a\,c-b\,c\,x]\,\text{PolyLog}[2,\,c\,(a+b\,x)]}{e}+\frac{b}{e}\int\frac{\text{Log}[1-a\,c-b\,c\,x]^2}{a+b\,x}\,dx$$

Program code:

```
Int[PolyLog[2,c_.*(a_.+b_.*x_)]/(d_.+e_.*x_),x_Symbol] :=
  Log[1-a*c-b*c*x]*PolyLog[2,c*(a+b*x)]/e + b/e*Int[Log[1-a*c-b*c*x]^2/(a+b*x),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c*(b*d-a*e)+e,0]
```

2: $\int \frac{\text{PolyLog}[2,\, c\,(a+b\,x)]}{d+e\,x}\,dx$ when $c\,(b\,d-a\,e)+e\neq 0$

Derivation: Integration by parts

- **Basis:** $\partial_x \text{PolyLog}[2,\, c\,(a+b\,x)] = -\frac{b\,\text{Log}[1-a\,c-b\,c\,x]}{a+b\,x}$

- **Rule: If** $c\,(b\,d-a\,e)+e\neq 0$**, then**

$$\int \frac{\text{PolyLog}[2,\, c\,(a+b\,x)]}{d+e\,x}\,dx \longrightarrow \frac{\text{Log}[d+e\,x]\,\text{PolyLog}[2,\, c\,(a+b\,x)]}{e}+\frac{b}{e}\int \frac{\text{Log}[d+e\,x]\,\text{Log}[1-a\,c-b\,c\,x]}{a+b\,x}\,dx$$

- **Program code:**

```
Int[PolyLog[2,c_.*(a_.+b_.*x_)]/(d_.+e_.*x_),x_Symbol] :=
  Log[d+e*x]*PolyLog[2,c*(a+b*x)]/e + b/e*Int[Log[d+e*x]*Log[1-a*c-b*c*x]/(a+b*x),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[c*(b*d-a*e)+e,0]
```

2: $\int (d+e\,x)^m\,\text{PolyLog}[2,\, c\,(a+b\,x)]\,dx$ when $m\neq -1$

Derivation: Integration by parts

Rule: If $m\neq -1$**, then**

$$\int (d+e\,x)^m\,\text{PolyLog}[2,\, c\,(a+b\,x)]\,dx \longrightarrow \frac{(d+e\,x)^{m+1}\,\text{PolyLog}[2,\, c\,(a+b\,x)]}{e\,(m+1)}+\frac{b}{e\,(m+1)}\int \frac{(d+e\,x)^{m+1}\,\text{Log}[1-a\,c-b\,c\,x]}{a+b\,x}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*PolyLog[2,c_.*(a_.+b_.*x_)],x_Symbol] :=
  (d+e*x)^(m+1)*PolyLog[2,c*(a+b*x)]/(e*(m+1)) + b/(e*(m+1))*Int[(d+e*x)^(m+1)*Log[1-a*c-b*c*x]/(a+b*x),x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[m,-1]
```

x: $\int (d+e\,x)^m\,\text{PolyLog}[n,\,c\,(a+b\,x)^p]\,dx$ when $n>0\,\wedge\,m\in\mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $n>0\,\wedge\,m\in\mathbb{Z}^+$, then

$$\int (d+e\,x)^m\,\text{PolyLog}[n,\,c\,(a+b\,x)^p]\,dx\,\longrightarrow\,\frac{(d+e\,x)^{m+1}\,\text{PolyLog}[n,\,c\,(a+b\,x)^p]}{e\,(m+1)}-\frac{b\,p}{e\,(m+1)}\int\frac{(d+e\,x)^{m+1}\,\text{PolyLog}[n-1,\,c\,(a+b\,x)^p]}{a+b\,x}\,dx$$

Program code:

```
(* Int[(d_.+e_.*x_)^m_.*PolyLog[n_,c_.*(a_.+b_.*x_)^p_.],x_Symbol] :=
  (d+e*x)^(m+1)*PolyLog[n,c*(a+b*x)^p]/(e*(m+1)) -
  b*p/(e*(m+1))*Int[(d+e*x)^(m+1)*PolyLog[n-1,c*(a+b*x)^p]/(a+b*x),x] /;
FreeQ[{a,b,c,d,e,m,p},x] && GtQ[n,0] && IGtQ[m,0] *)
```

2: $\int x^m\,\text{PolyLog}[n,\,c\,(a+b\,x)^p]\,dx$ when $n>0\,\wedge\,m\in\mathbb{Z}\,\wedge\,m\neq-1$

Derivation: Integration by parts

Rule: If $n>0\,\wedge\,m\in\mathbb{Z}\,\wedge\,m\neq-1$, then

$$\int x^m\,\text{PolyLog}[n,\,c\,(a+b\,x)^p]\,dx\,\longrightarrow$$

$$-\frac{\left(a^{m+1}-b^{m+1}\,x^{m+1}\right)\,\text{PolyLog}[n,\,c\,(a+b\,x)^p]}{(m+1)\,b^{m+1}}+\frac{p}{(m+1)\,b^m}\int\text{PolyLog}[n-1,\,c\,(a+b\,x)^p]\,\text{ExpandIntegrand}\left[\frac{a^{m+1}-b^{m+1}\,x^{m+1}}{a+b\,x},\,x\right]dx$$

Program code:

```
Int[x_^m_.*PolyLog[n_,c_.*(a_.+b_.*x_)^p_.],x_Symbol] :=
  -(a^(m+1)-b^(m+1)*x^(m+1))*PolyLog[n,c*(a+b*x)^p]/((m+1)*b^(m+1)) +
  p/((m+1)*b^m)*Int[ExpandIntegrand[PolyLog[n-1,c*(a+b*x)^p],(a^(m+1)-b^(m+1)*x^(m+1))/(a+b*x),x],x] /;
FreeQ[{a,b,c,p},x] && GtQ[n,0] && IntegerQ[m] && NeQ[m,-1]
```

3. $\int u\,(g+h\,\text{Log}[f\,(d+e\,x)^n])\,\text{PolyLog}[2,\,c\,(a+b\,x)]\,dx$

1: $\int (g+h\,\text{Log}[f\,(d+e\,x)^n])\,\text{PolyLog}[2,\,c\,(a+b\,x)]\,dx$

Derivation: Integration by parts and algebraic expansion

- **Basis:** $\partial_x\left((g+h\,\text{Log}[f\,(d+e\,x)^n])\,\text{PolyLog}[2,\,c\,(a+b\,x)]\right) == -\frac{b\,(g+h\,\text{Log}[f\,(d+e\,x)^n])\,\text{Log}[1-c\,(a+b\,x)]}{a+b\,x}+\frac{e\,h\,n\,\text{PolyLog}[2,c\,(a+b\,x)]}{d+e\,x}$

Rule:

$$\int (g+h\,\text{Log}[f\,(d+e\,x)^n])\,\text{PolyLog}[2,\,c\,(a+b\,x)]\,dx \longrightarrow$$

$$x\,(g+h\,\text{Log}[f\,(d+e\,x)^n])\,\text{PolyLog}[2,\,c\,(a+b\,x)]+$$
$$b\int (g+h\,\text{Log}[f\,(d+e\,x)^n])\,\text{Log}[1-a\,c-b\,c\,x]\,\text{ExpandIntegrand}\left[\frac{x}{a+b\,x},\,x\right]dx-e\,h\,n\int \text{PolyLog}[2,\,c\,(a+b\,x)]\,\text{ExpandIntegrand}\left[\frac{x}{d+e\,x},\,x\right]dx$$

Program code:

```
Int[(g_.+h_.*Log[f_.*(d_.+e_.*x_)^n_.])*PolyLog[2,c_.*(a_.+b_.*x_)],x_Symbol] :=
  x*(g+h*Log[f*(d+e*x)^n])*PolyLog[2,c*(a+b*x)] +
  b*Int[(g+h*Log[f*(d+e*x)^n])*Log[1-a*c-b*c*x]*ExpandIntegrand[x/(a+b*x),x],x] -
  e*h*n*Int[PolyLog[2,c*(a+b*x)]*ExpandIntegrand[x/(d+e*x),x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,n},x]
```

2. $\int x^m\,(g+h\,\text{Log}[f\,(d+e\,x)^n])\,\text{PolyLog}[2,\,c\,(a+b\,x)]\,dx$ when $m\in\mathbb{Z}$

1. $\int \frac{(g+h\,\text{Log}[1+e\,x])\,\text{PolyLog}[2,\,c\,x]}{x}\,dx$ when $c+e==0$

1: $\int \frac{\text{Log}[1+e\,x]\,\text{PolyLog}[2,\,c\,x]}{x}\,dx$ when $c+e==0$

Derivation: Integration by substitution

Basis: If $e+c==0$**, then** $\frac{\text{Log}[1+e\,x]}{x}==-\partial_x\text{PolyLog}[2,\,c\,x]$

Rule: If $c+e==0$**, then**

$$\int \frac{\text{Log}[1+e\,x]\,\text{PolyLog}[2,\,c\,x]}{x}\,dx \longrightarrow -\frac{\text{PolyLog}[2,\,c\,x]^2}{2}$$

Program code:

```
Int[Log[1+e_.*x_]*PolyLog[2,c_.*x_]/x_,x_Symbol] :=
  -PolyLog[2,c*x]^2/2 /;
FreeQ[{c,e},x] && EqQ[c+e,0]
```

2: $\int \frac{(g+h\,\text{Log}[1+e\,x])\,\text{PolyLog}[2,\,c\,x]}{x}\,dx$ when $c+e==0$

Derivation: Algebraic expansion

Rule: If $c+e==0$**, then**

$$\int \frac{(g+h\,\text{Log}[1+e\,x])\,\text{PolyLog}[2,\,c\,x]}{x}\,dx \longrightarrow g\int \frac{\text{PolyLog}[2,\,c\,x]}{x}\,dx+h\int \frac{\text{Log}[1+e\,x]\,\text{PolyLog}[2,\,c\,x]}{x}\,dx$$

Program code:

```
Int[(g_+h_.*Log[1+e_.*x_])*PolyLog[2,c_.*x_]/x_,x_Symbol] :=
  g*Int[PolyLog[2,c*x]/x,x] + h*Int[(Log[1+e*x]*PolyLog[2,c*x])/x,x] /;
FreeQ[{c,e,g,h},x] && EqQ[c+e,0]
```

2: $\int x^m \left(g+h \operatorname{Log}\left[f\left(d+e x\right)^n\right]\right) \operatorname{PolyLog}[2, c (a+b x)] \, dx$ when $m \in \mathbb{Z} \wedge m \neq -1$

Derivation: Integration by parts and algebraic expansion

- **Basis:** $\partial_x \left(\left(g+h \operatorname{Log}\left[f (d+e x)^n\right]\right) \operatorname{PolyLog}[2, c (a+b x)]\right) = -\frac{b \left(g+h \operatorname{Log}\left[f (d+e x)^n\right]\right) \operatorname{Log}[1-a c-b c x]}{a+b x}+\frac{e h n \operatorname{PolyLog}[2, c (a+b x)]}{d+e x}$

- **Rule: If** $m \in \mathbb{Z} \wedge m \neq -1$, **then**

$$\int x^m \left(g+h \operatorname{Log}\left[f (d+e x)^n\right]\right) \operatorname{PolyLog}[2, c (a+b x)] \, dx \longrightarrow$$

$$\frac{x^{m+1} \left(g+h \operatorname{Log}\left[f (d+e x)^n\right]\right) \operatorname{PolyLog}[2, c (a+b x)]}{m+1}+$$

$$\frac{b}{m+1} \int \left(g+h \operatorname{Log}\left[f (d+e x)^n\right]\right) \operatorname{Log}[1-a c-b c x] \operatorname{ExpandIntegrand}\left[\frac{x^{m+1}}{a+b x}, x\right] dx-\frac{e h n}{m+1} \int \operatorname{PolyLog}[2, c (a+b x)] \operatorname{ExpandIntegrand}\left[\frac{x^{m+1}}{d+e x}, x\right] dx$$

Program code:

```
Int[x_^m_.*(g_.+h_.*Log[f_.*(d_.+e_.*x_)^n_.])*PolyLog[2,c_.*(a_.+b_.*x_)],x_Symbol] :=
  x^(m+1)*(g+h*Log[f*(d+e*x)^n])*PolyLog[2,c*(a+b*x)]/(m+1) +
  b/(m+1)*Int[ExpandIntegrand[(g+h*Log[f*(d+e*x)^n])*Log[1-a*c-b*c*x],x^(m+1)/(a+b*x),x],x] -
  e*h*n/(m+1)*Int[ExpandIntegrand[PolyLog[2,c*(a+b*x)],x^(m+1)/(d+e*x),x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,n},x] && IntegerQ[m] && NeQ[m,-1]
```

3: $\int P[x]\,(g+h\,\text{Log}[f\,(d+e\,x)^n])\,\text{PolyLog}[2,\,c\,(a+b\,x)]\,dx$

Derivation: Integration by parts and algebraic expansion

- **Basis:** $\partial_x\left((g+h\,\text{Log}[f\,(d+e\,x)^n])\,\text{PolyLog}[2,\,c\,(a+b\,x)]\right) = -\frac{b\,\left(g+h\,\text{Log}\left[f\,(d+e\,x)^n\right]\right)\,\text{Log}[1-a\,c-b\,c\,x]}{a+b\,x}+\frac{e\,h\,n\,\text{PolyLog}[2,c\,(a+b\,x)]}{d+e\,x}$

- **Rule:** Let $u\to\int P[x]\,dx$, then

$$\int P[x]\,(g+h\,\text{Log}[f\,(d+e\,x)^n])\,\text{PolyLog}[2,\,c\,(a+b\,x)]\,dx\longrightarrow$$

$$u\,(g+h\,\text{Log}[f\,(d+e\,x)^n])\,\text{PolyLog}[2,\,c\,(a+b\,x)]+$$
$$b\int(g+h\,\text{Log}[f\,(d+e\,x)^n])\,\text{Log}[1-a\,c-b\,c\,x]\,\text{ExpandIntegrand}\left[\frac{u}{a+b\,x},\,x\right]dx-e\,h\,n\int\text{PolyLog}[2,\,c\,(a+b\,x)]\,\text{ExpandIntegrand}\left[\frac{u}{d+e\,x},\,x\right]dx$$

Program code:

```
Int[Px_*(g_.+h_.*Log[f_.*(d_.+e_.*x_)^n_.])*PolyLog[2,c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{u=IntHide[Px,x]},
  u*(g+h*Log[f*(d+e*x)^n])*PolyLog[2,c*(a+b*x)] +
  b*Int[ExpandIntegrand[(g+h*Log[f*(d+e*x)^n])*Log[1-a*c-b*c*x],u/(a+b*x),x],x] -
  e*h*n*Int[ExpandIntegrand[PolyLog[2,c*(a+b*x)],u/(d+e*x),x],x]] /;
FreeQ[{a,b,c,d,e,f,g,h,n},x] && PolyQ[Px,x]
```

4. $\int x^m P[x] (g+h \text{Log}[f (d+e x)^n]) \text{PolyLog}[2, c (a+b x)] \, dx$ when $m \in \mathbb{Z}$

1: $\int x^m P[x] (g+h \text{Log}[1+e x]) \text{PolyLog}[2, c x] \, dx$ when $m \in \mathbb{Z}^- \land c+e == 0 \land P[x, -m-1] \neq 0$

Derivation: Algebraic expansion

Note: Separates out the term in the integrand of the form $\frac{P[x,-m-1] (g+h \text{Log}[1+e x]) \text{PolyLog}[2,c x]}{x}$.

Rule: If $m \in \mathbb{Z}^- \land c+e == 0 \land P[x, -m-1] \neq 0$**, then**

$$\int x^m P[x] (g+h \text{Log}[1+e x]) \text{PolyLog}[2, c x] \, dx \longrightarrow$$

$$P[x, -m-1] \int \frac{(g+h \text{Log}[1+e x]) \text{PolyLog}[2, c x]}{x} \, dx + \int x^m \left(P[x] - P[x, -m-1] \, x^{-m-1}\right) (g+h \text{Log}[1+e x]) \text{PolyLog}[2, c x] \, dx$$

Program code:

```
Int[x_^m_*Px_*(g_.+h_.*Log[1+e_.*x_])*PolyLog[2,c_.*x_],x_Symbol] :=
  Coeff[Px,x,-m-1]*Int[(g+h*Log[1+e*x])*PolyLog[2,c*x]/x,x] +
  Int[x^m*(Px-Coeff[Px,x,-m-1]*x^(-m-1))*(g+h*Log[1+e*x])*PolyLog[2,c*x],x] /;
FreeQ[{c,e,g,h},x] && PolyQ[Px,x] && ILtQ[m,0] && EqQ[c+e,0] && NeQ[Coeff[Px,x,-m-1],0]
```

2: $\int x^m P[x] (g + h \operatorname{Log}[f (d + e x)^n]) \operatorname{PolyLog}[2, c (a + b x)] \, dx$ when $m \in \mathbb{Z}$

Derivation: Integration by parts and algebraic expansion

- **Basis:** $\partial_x ((g + h \operatorname{Log}[f (d + e x)^n]) \operatorname{PolyLog}[2, c (a + b x)]) == -\frac{b (g+h \operatorname{Log}[f (d+e x)^n]) \operatorname{Log}[1-a c-b c x]}{a+b x} + \frac{e h n \operatorname{PolyLog}[2,c (a+b x)]}{d+e x}$

Rule: If $m \in \mathbb{Z}$, **let** $u \to \int x^m P[x] \, dx$, **then**

$$\int x^m P[x] (g + h \operatorname{Log}[f (d + e x)^n]) \operatorname{PolyLog}[2, c (a + b x)] \, dx \longrightarrow$$

$$u (g + h \operatorname{Log}[f (d + e x)^n]) \operatorname{PolyLog}[2, c (a + b x)] + \\ b \int (g + h \operatorname{Log}[f (d + e x)^n]) \operatorname{Log}[1 - a c - b c x] \operatorname{ExpandIntegrand}\left[\frac{u}{a + b x}, x\right] dx - e h n \int \operatorname{PolyLog}[2, c (a + b x)] \operatorname{ExpandIntegrand}\left[\frac{u}{d + e x}, x\right] dx$$

Program code:

```
Int[x_^m_.*Px_*(g_.+h_.*Log[f_.*(d_.+e_.*x_)^n_.])*PolyLog[2,c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{u=IntHide[x^m*Px,x]},
  u*(g+h*Log[f*(d+e*x)^n])*PolyLog[2,c*(a+b*x)] +
  b*Int[ExpandIntegrand[(g+h*Log[f*(d+e*x)^n])*Log[1-a*c-b*c*x],u/(a+b*x),x],x] -
  e*h*n*Int[ExpandIntegrand[PolyLog[2,c*(a+b*x)],u/(d+e*x),x],x]] /;
FreeQ[{a,b,c,d,e,f,g,h,n},x] && PolyQ[Px,x] && IntegerQ[m]
```

U: $\int x^m P[x] (g + h \operatorname{Log}[f (d + e x)^n]) \operatorname{PolyLog}[2, c (a + b x)] \, dx$

Rule:

$$\int x^m P[x] (g + h \operatorname{Log}[f (d + e x)^n]) \operatorname{PolyLog}[2, c (a + b x)] \, dx \longrightarrow \int x^m P[x] (g + h \operatorname{Log}[f (d + e x)^n]) \operatorname{PolyLog}[2, c (a + b x)] \, dx$$

Program code:

```
Int[x_^m_*Px_.*(g_.+h_.*Log[f_.*(d_.+e_.*x_)^n_.])*PolyLog[2,c_.*(a_.+b_.*x_)],x_Symbol] :=
  Unintegrable[x^m*Px*(g+h*Log[f*(d+e*x)^n])*PolyLog[2,c*(a+b*x)],x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && PolyQ[Px,x]
```

4. $\int u\,\text{PolyLog}\left[n,\ d\left(F^{c\,(a+b\,x)}\right)^p\right]\,dx$

1: $\int \text{PolyLog}\left[n,\ d\left(F^{c\,(a+b\,x)}\right)^p\right]\,dx$

Derivation: Primitive rule

- **Basis:** $\partial_z\,\text{PolyLog}[n,\ z] = \frac{\text{PolyLog}[n-1,z]}{z}$

Rule:

$$\int \text{PolyLog}\left[n,\ d\left(F^{c\,(a+b\,x)}\right)^p\right]\,dx \longrightarrow \frac{\text{PolyLog}\left[n+1,\ d\left(F^{c\,(a+b\,x)}\right)^p\right]}{b\,c\,p\,\text{Log}[F]}$$

Program code:

```
Int[PolyLog[n_,d_.*(F_^(c_.*(a_.+b_.*x_)))^p_.],x_Symbol] :=
  PolyLog[n+1,d*(F^(c*(a+b*x)))^p]/(b*c*p*Log[F]) /;
FreeQ[{F,a,b,c,d,n,p},x]
```

2: $\int (e+f\,x)^m\,\text{PolyLog}\left[n,\ d\left(F^{c\,(a+b\,x)}\right)^p\right]\,dx$ when $m>0$

Derivation: Integration by parts

- **Basis:** $\text{PolyLog}\left[n,\ d\left(F^{c\,(a+b\,x)}\right)^p\right] = \partial_x\,\frac{\text{PolyLog}\left[n+1,d\left(F^{c\,(a+b\,x)}\right)^p\right]}{b\,c\,p\,\text{Log}[F]}$

Rule: If $m>0$**, then**

$$\int (e+f\,x)^m\,\text{PolyLog}\left[n,\ d\left(F^{c\,(a+b\,x)}\right)^p\right]\,dx \longrightarrow$$

$$\frac{(e+f\,x)^m\,\text{PolyLog}\left[n+1,\ d\left(F^{c\,(a+b\,x)}\right)^p\right]}{b\,c\,p\,\text{Log}[F]} - \frac{f\,m}{b\,c\,p\,\text{Log}[F]}\int (e+f\,x)^{m-1}\,\text{PolyLog}\left[n+1,\ d\left(F^{c\,(a+b\,x)}\right)^p\right]\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*PolyLog[n_,d_.*(F_^(c_.*(a_.+b_.*x_)))^p_.],x_Symbol] :=
  (e+f*x)^m*PolyLog[n+1,d*(F^(c*(a+b*x)))^p]/(b*c*p*Log[F]) -
  f*m/(b*c*p*Log[F])*Int[(e+f*x)^(m-1)*PolyLog[n+1,d*(F^(c*(a+b*x)))^p],x] /;
FreeQ[{F,a,b,c,d,e,f,n,p},x] && GtQ[m,0]
```

5. $\int u\,\frac{\text{PolyLog}[n,\,F[x]]\,F'[x]}{F[x]}\,dx$

1: $\int \frac{\text{PolyLog}[n,\,F[x]]\,F'[x]}{F[x]}\,dx$

- **Basis:** $\partial_x \text{PolyLog}[n+1,\,x] == \frac{\text{PolyLog}[n,x]}{x}$
- **Rule:**

$$\int \frac{\text{PolyLog}[n,\,F[x]]\,F'[x]}{F[x]}\,dx \longrightarrow \text{PolyLog}[n+1,\,F[x]]$$

- **Program code:**

```
Int[u_*PolyLog[n_,v_],x_Symbol] :=
  With[{w=DerivativeDivides[v,u*v,x]},
  w*PolyLog[n+1,v] /;
 Not[FalseQ[w]]] /;
FreeQ[n,x]
```

2: $\int \frac{\text{Log}[G[x]]\,\text{PolyLog}[n,\,F[x]]\,F'[x]}{F[x]}\,dx$

- **Derivation: Integration by parts**
- **Basis:** $\frac{\text{PolyLog}[n,x]}{x} == \partial_x \text{PolyLog}[n+1,\,x]$
- **Rule:**

$$\int \frac{\text{Log}[G[x]]\,\text{PolyLog}[n,\,F[x]]\,F'[x]}{F[x]}\,dx \longrightarrow \text{Log}[G[x]]\,\text{PolyLog}[n+1,\,F[x]] - \int \frac{G'[x]\,\text{PolyLog}[n+1,\,F[x]]}{G[x]}\,dx$$

- **Program code:**

```
Int[u_*Log[w_]*PolyLog[n_,v_],x_Symbol] :=
  With[{z=DerivativeDivides[v,u*v,x]},
  z*Log[w]*PolyLog[n+1,v] -
  Int[SimplifyIntegrand[z*D[w,x]*PolyLog[n+1,v]/w,x],x] /;
 Not[FalseQ[z]]] /;
FreeQ[n,x] && InverseFunctionFreeQ[w,x]
```