Rules for integrands of the form $(c + d x)^m (a + b \operatorname{Sin}[e + f x])^n$

1. $\int (c + d x)^m (b \operatorname{Sin}[e + f x])^n \, dx$

   1. $\int (c + d x)^m (b \operatorname{Sin}[e + f x])^n \, dx$ when $n > 0$

      1. $\int (c + d x)^m \operatorname{Sin}[e + f x] \, dx$

         1: $\int (c + d x)^m \operatorname{Sin}[e + f x] \, dx$ when $m > 0$

- **Reference: CRC 392, A&S 4.3.119**
- **Reference: CRC 396, A&S 4.3.123**
- **Derivation: Integration by parts**
- **Basis:** $\operatorname{Sin}[e + f x] == -\frac{1}{f} \partial_x \operatorname{Cos}[e + f x]$
- **Rule: If** $m > 0$, **then**

$$\int (c + d x)^m \operatorname{Sin}[e + f x] \, dx \longrightarrow -\frac{(c + d x)^m \operatorname{Cos}[e + f x]}{f} + \frac{d m}{f} \int (c + d x)^{m-1} \operatorname{Cos}[e + f x] \, dx$$

- **Program code:**

```
Int[(c_.+d_.*x_)^m_.*sin[e_.+f_.*x_],x_Symbol] :=
  -(c+d*x)^m*Cos[e+f*x]/f +
  d*m/f*Int[(c+d*x)^(m-1)*Cos[e+f*x],x] /;
FreeQ[{c,d,e,f},x] && GtQ[m,0]
```

2: $\int (c+d\,x)^m \operatorname{Sin}[e+f\,x]\,dx$ when $m < -1$

- **Reference: CRC 405, A&S 4.3.120**
- **Reference: CRC 406, A&S 4.3.124**
- **Derivation: Integration by parts**
- **Rule: If** $m < -1$**, then**

$$\int (c+d\,x)^m \operatorname{Sin}[e+f\,x]\,dx \longrightarrow \frac{(c+d\,x)^{m+1}\operatorname{Sin}[e+f\,x]}{d\,(m+1)} - \frac{f}{d\,(m+1)}\int (c+d\,x)^{m+1}\operatorname{Cos}[e+f\,x]\,dx$$

- **Program code:**

```
Int[(c_.+d_.*x_)^m_*sin[e_.+f_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*Sin[e+f*x]/(d*(m+1)) -
  f/(d*(m+1))*Int[(c+d*x)^(m+1)*Cos[e+f*x],x] /;
FreeQ[{c,d,e,f},x] && LtQ[m,-1]
```

3. $\int \frac{\operatorname{Sin}[e+f\,x]}{c+d\,x}\,dx$

1: $\int \frac{\operatorname{Sin}[e+f\,x]}{c+d\,x}\,dx$ when $d\,e-c\,f == 0$

Derivation: Primitive rule

Basis: $\operatorname{SinIntegral}[i\,z] == i\,\operatorname{SinhIntegral}[z]$

Basis: $\partial_x \operatorname{CosIntegral}[i\,F[x]] == \partial_x \operatorname{CoshIntegral}[F[x]] == \partial_x \operatorname{CoshIntegral}[-F[x]]$

Rule: If $d\,e-c\,f == 0$**, then**

$$\int \frac{\operatorname{Sin}[e+f\,x]}{c+d\,x}\,dx \longrightarrow \frac{\operatorname{SinIntegral}[e+f\,x]}{d}$$

$$\int \frac{\operatorname{Cos}[e+f\,x]}{c+d\,x}\,dx \longrightarrow \frac{\operatorname{CosIntegral}[e+f\,x]}{d}$$

Program code:

```
Int[sin[e_.+f_.*Complex[0,fz_]*x_]/(c_.+d_.*x_),x_Symbol] :=
  I*SinhIntegral[c*f*fz/d+f*fz*x]/d /;
FreeQ[{c,d,e,f,fz},x] && EqQ[d*e-c*f*fz*I,0]
```

```
Int[sin[e_.+f_.*x_]/(c_.+d_.*x_),x_Symbol] :=
  SinIntegral[e+f*x]/d /;
FreeQ[{c,d,e,f},x] && EqQ[d*e-c*f,0]
```

```
Int[sin[e_.+f_.*Complex[0,fz_]*x_]/(c_.+d_.*x_),x_Symbol] :=
  CoshIntegral[-c*f*fz/d-f*fz*x]/d /;
FreeQ[{c,d,e,f,fz},x] && EqQ[d*(e-Pi/2)-c*f*fz*I,0] && NegQ[c*f*fz/d,0]
```

```
Int[sin[e_.+f_.*Complex[0,fz_]*x_]/(c_.+d_.*x_),x_Symbol] :=
  CoshIntegral[c*f*fz/d+f*fz*x]/d /;
FreeQ[{c,d,e,f,fz},x] && EqQ[d*(e-Pi/2)-c*f*fz*I,0]
```

```
Int[sin[e_.+f_.*x_]/(c_.+d_.*x_),x_Symbol] :=
  CosIntegral[e-Pi/2+f*x]/d /;
FreeQ[{c,d,e,f},x] && EqQ[d*(e-Pi/2)-c*f,0]
```

2: $\int \frac{\mathrm{Sin}[e+f\,x]}{c+d\,x}\,dx$ when $d\,e-c\,f\neq 0$

Derivation: Algebraic expansion

Basis: $\mathrm{Sin}[e+f\,x] == \mathrm{Cos}\left[\frac{d\,e-c\,f}{d}\right]\mathrm{Sin}\left[\frac{c\,f}{d}+f\,x\right]+\mathrm{Sin}\left[\frac{d\,e-c\,f}{d}\right]\mathrm{Cos}\left[\frac{c\,f}{d}+f\,x\right]$

Rule: If $d\,e-c\,f\neq 0$, **then**

$$\int \frac{\mathrm{Sin}[e+f\,x]}{c+d\,x}\,dx \longrightarrow \mathrm{Cos}\left[\frac{d\,e-c\,f}{d}\right]\int \frac{\mathrm{Sin}\left[\frac{c\,f}{d}+f\,x\right]}{c+d\,x}\,dx+\mathrm{Sin}\left[\frac{d\,e-c\,f}{d}\right]\int \frac{\mathrm{Cos}\left[\frac{c\,f}{d}+f\,x\right]}{c+d\,x}\,dx$$

Program code:

```
Int[sin[e_.+f_.*x_]/(c_.+d_.*x_),x_Symbol] :=
  Cos[(d*e-c*f)/d]*Int[Sin[c*f/d+f*x]/(c+d*x),x] +
  Sin[(d*e-c*f)/d]*Int[Cos[c*f/d+f*x]/(c+d*x),x] /;
FreeQ[{c,d,e,f},x] && NeQ[d*e-c*f,0]
```

4. $\int \frac{\mathrm{Sin}[e+f\,x]}{\sqrt{c+d\,x}}\,dx$

1: $\int \frac{\mathrm{Sin}[e+f\,x]}{\sqrt{c+d\,x}}\,dx$ when $d\,e-c\,f == 0$

Derivation: Integration by substitution

Basis: If $d\,e-c\,f == 0$, **then** $\frac{F[e+f\,x]}{\sqrt{c+d\,x}} == \frac{2}{d}\,\mathrm{Subst}\left[F\left[\frac{f\,x^2}{d}\right],\,x,\,\sqrt{c+d\,x}\right]\,\partial_x\sqrt{c+d\,x}$

Rule: If $d\,e-c\,f == 0$, **then**

$$\int \frac{\mathrm{Sin}[e+f\,x]}{\sqrt{c+d\,x}}\,dx \longrightarrow \frac{2}{d}\,\mathrm{Subst}\left[\int \mathrm{Sin}\left[\frac{f\,x^2}{d}\right]dx,\,x,\,\sqrt{c+d\,x}\right]$$

Program code:

```
Int[sin[e_.+Pi/2+f_.*x_]/Sqrt[c_.+d_.*x_],x_Symbol] :=
  2/d*Subst[Int[Cos[f*x^2/d],x],x,Sqrt[c+d*x]] /;
FreeQ[{c,d,e,f},x] && ComplexFreeQ[f] && EqQ[d*e-c*f,0]
```

```
Int[sin[e_.+f_.*x_]/Sqrt[c_.+d_.*x_],x_Symbol] :=
  2/d*Subst[Int[Sin[f*x^2/d],x],x,Sqrt[c+d*x]] /;
FreeQ[{c,d,e,f},x] && ComplexFreeQ[f] && EqQ[d*e-c*f,0]
```

2: $\int \frac{\mathrm{Sin}[e+f\,x]}{\sqrt{c+d\,x}}\,dx$ when $d\,e-c\,f \neq 0$

Derivation: Algebraic expansion

Basis: $\mathrm{Sin}[e+f\,x] == \mathrm{Cos}\left[\frac{d\,e-c\,f}{d}\right]\,\mathrm{Sin}\left[\frac{c\,f}{d}+f\,x\right]+\mathrm{Sin}\left[\frac{d\,e-c\,f}{d}\right]\,\mathrm{Cos}\left[\frac{c\,f}{d}+f\,x\right]$

Rule: If $d\,e-c\,f \neq 0$, **then**

$$\int \frac{\mathrm{Sin}[e+f\,x]}{\sqrt{c+d\,x}}\,dx \longrightarrow \mathrm{Cos}\left[\frac{d\,e-c\,f}{d}\right]\int \frac{\mathrm{Sin}\left[\frac{c\,f}{d}+f\,x\right]}{\sqrt{c+d\,x}}\,dx+\mathrm{Sin}\left[\frac{d\,e-c\,f}{d}\right]\int \frac{\mathrm{Cos}\left[\frac{c\,f}{d}+f\,x\right]}{\sqrt{c+d\,x}}\,dx$$

Program code:

```
Int[sin[e_.+f_.*x_]/Sqrt[c_.+d_.*x_],x_Symbol] :=
  Cos[(d*e-c*f)/d]*Int[Sin[c*f/d+f*x]/Sqrt[c+d*x],x] +
  Sin[(d*e-c*f)/d]*Int[Cos[c*f/d+f*x]/Sqrt[c+d*x],x] /;
FreeQ[{c,d,e,f},x] && ComplexFreeQ[f] && NeQ[d*e-c*f,0]
```

5: $\int (c+d\,x)^m \operatorname{Sin}[e+f\,x]\,dx$

Derivation: Algebraic expansion

- Basis: $\operatorname{Sin}[z] == \frac{1}{2}\,i\,e^{-i\,z} - \frac{1}{2}\,i\,e^{i\,z}$
- Basis: $\operatorname{Cos}[z] == \frac{1}{2}\,e^{-i\,z} + \frac{1}{2}\,e^{i\,z}$

Rule:

$$\int (c+d\,x)^m \operatorname{Sin}[e+f\,x]\,dx \longrightarrow \frac{i}{2}\int (c+d\,x)^m e^{-i\,(e+f\,x)}\,dx - \frac{i}{2}\int (c+d\,x)^m e^{i\,(e+f\,x)}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*sin[e_.+k_.*Pi+f_.*x_],x_Symbol] :=
  I/2*Int[(c+d*x)^m*E^(-I*k*Pi)*E^(-I*(e+f*x)),x] - I/2*Int[(c+d*x)^m*E^(I*k*Pi)*E^(I*(e+f*x)),x] /;
FreeQ[{c,d,e,f,m},x] && IntegerQ[2*k]
```

```
Int[(c_.+d_.*x_)^m_.*sin[e_.+f_.*x_],x_Symbol] :=
  I/2*Int[(c+d*x)^m*E^(-I*(e+f*x)),x] - I/2*Int[(c+d*x)^m*E^(I*(e+f*x)),x] /;
FreeQ[{c,d,e,f,m},x]
```

2. $\int (c+d\,x)^m\,(b\operatorname{Sin}[e+f\,x])^n\,dx$ when $n>1$

1: $\int (c+d\,x)^m \operatorname{Sin}[e+f\,x]^2\,dx$

Derivation: Algebraic expansion

- Basis: $\operatorname{Sin}[z]^2 == \frac{1}{2} - \frac{\operatorname{Cos}[2\,z]}{2}$

Rule:

$$\int (c+d\,x)^m \operatorname{Sin}[e+f\,x]^2\,dx \longrightarrow \frac{1}{2}\int (c+d\,x)^m\,dx - \frac{1}{2}\int (c+d\,x)^m \operatorname{Cos}[2\,e+2\,f\,x]\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*sin[e_.+f_.*x_/2]^2,x_Symbol] :=
  1/2*Int[(c+d*x)^m,x] - 1/2*Int[(c+d*x)^m*Cos[2*e+f*x],x] /;
FreeQ[{c,d,e,f,m},x]
```

2. $\int (c+d\,x)^m\,(b\,\text{Sin}[e+f\,x])^n\,dx$ when $n>1\wedge m\geq 1$

1: $\int (c+d\,x)\,(b\,\text{Sin}[e+f\,x])^n\,dx$ when $n>1$

Reference: G&R 2.631.2 with $m\to 1$

Reference: G&R 2.631.3 with $m\to 1$

Rule: If $n>1$, then

$$\int (c+d\,x)\,(b\,\text{Sin}[e+f\,x])^n\,dx \longrightarrow$$

$$\frac{d\,(b\,\text{Sin}[e+f\,x])^n}{f^2\,n^2}-\frac{b\,(c+d\,x)\,\text{Cos}[e+f\,x]\,(b\,\text{Sin}[e+f\,x])^{n-1}}{f\,n}+\frac{b^2\,(n-1)}{n}\int (c+d\,x)\,(b\,\text{Sin}[e+f\,x])^{n-2}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)*(b_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  d*(b*Sin[e+f*x])^n/(f^2*n^2) -
  b*(c+d*x)*Cos[e+f*x]*(b*Sin[e+f*x])^(n-1)/(f*n) +
  b^2*(n-1)/n*Int[(c+d*x)*(b*Sin[e+f*x])^(n-2),x] /;
FreeQ[{b,c,d,e,f},x] && GtQ[n,1]
```

2: $\int (c+d\,x)^m\,(b\,\text{Sin}[e+f\,x])^n\,dx$ when $n>1\,\wedge\,m>1$

Reference: G&R 2.631.2

Reference: G&R 2.631.3

Rule: If $n>1\,\wedge\,m>1$, then

$$\int (c+d\,x)^m\,(b\,\text{Sin}[e+f\,x])^n\,dx \longrightarrow$$
$$\frac{d\,m\,(c+d\,x)^{m-1}\,(b\,\text{Sin}[e+f\,x])^n}{f^2\,n^2}-\frac{b\,(c+d\,x)^m\,\text{Cos}[e+f\,x]\,(b\,\text{Sin}[e+f\,x])^{n-1}}{f\,n}+$$
$$\frac{b^2\,(n-1)}{n}\int (c+d\,x)^m\,(b\,\text{Sin}[e+f\,x])^{n-2}\,dx-\frac{d^2\,m\,(m-1)}{f^2\,n^2}\int (c+d\,x)^{m-2}\,(b\,\text{Sin}[e+f\,x])^n\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_*(b_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  d*m*(c+d*x)^(m-1)*(b*Sin[e+f*x])^n/(f^2*n^2) -
  b*(c+d*x)^m*Cos[e+f*x]*(b*Sin[e+f*x])^(n-1)/(f*n) +
  b^2*(n-1)/n*Int[(c+d*x)^m*(b*Sin[e+f*x])^(n-2),x] -
  d^2*m*(m-1)/(f^2*n^2)*Int[(c+d*x)^(m-2)*(b*Sin[e+f*x])^n,x] /;
FreeQ[{b,c,d,e,f},x] && GtQ[n,1] && GtQ[m,1]
```

3. $\int (c+d\,x)^m\,(b\,\text{Sin}[e+f\,x])^n\,dx$ when $n>1\,\wedge\,m<1$

1: $\int (c+d\,x)^m\,\text{Sin}[e+f\,x]^n\,dx$ when $n\in\mathbb{Z}\,\wedge\,n>1\,\wedge\,-1\le m<1$

Derivation: Algebraic exnansion

Rule: If $n\in\mathbb{Z}\,\wedge\,n>1\,\wedge\,-1\le m<1$, then

$$\int (c+d\,x)^m\,\text{Sin}[e+f\,x]^n\,dx \longrightarrow \int (c+d\,x)^m\,\text{TrigReduce}[\text{Sin}[e+f\,x]^n]\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_*sin[e_.+f_.*x_]^n_,x_Symbol] :=
  Int[ExpandTrigReduce[(c+d*x)^m,Sin[e+f*x]^n,x],x] /;
FreeQ[{c,d,e,f,m},x] && IGtQ[n,1] && (Not[RationalQ[m]] || GeQ[m,-1] && LtQ[m,1])
```

2: $\int (c+d\,x)^m\,\mathrm{Sin}[e+f\,x]^n\,dx$ when $n\in\mathbb{Z}\wedge n>1\wedge -2\le m<-1$

Derivation: Integration by parts

Rule: If $n\in\mathbb{Z}\wedge n>1\wedge -2\le m<-1$, then

$$\int (c+d\,x)^m\,\mathrm{Sin}[e+f\,x]^n\,dx\ \longrightarrow\ \frac{(c+d\,x)^{m+1}\,\mathrm{Sin}[e+f\,x]^n}{d\,(m+1)}-\frac{f\,n}{d\,(m+1)}\int (c+d\,x)^{m+1}\,\mathrm{TrigReduce}\big[\mathrm{Cos}[e+f\,x]\,\mathrm{Sin}[e+f\,x]^{n-1}\big]\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_*sin[e_.+f_.*x_]^n_,x_Symbol] :=
  (c+d*x)^(m+1)*Sin[e+f*x]^n/(d*(m+1)) -
  f*n/(d*(m+1))*Int[ExpandTrigReduce[(c+d*x)^(m+1),Cos[e+f*x]*Sin[e+f*x]^(n-1),x],x] /;
FreeQ[{c,d,e,f,m},x] && IGtQ[n,1] && GeQ[m,-2] && LtQ[m,-1]
```

3: $\int (c+d\,x)^m\,(b\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $n>1\wedge m<-2$

Reference: G&R 2.638.1

Reference: G&R 2.638.2

Rule: If $n>1\wedge m<-2$, then

$$\int (c+d\,x)^m\,(b\,\mathrm{Sin}[e+f\,x])^n\,dx\ \longrightarrow$$

$$\frac{(c+d\,x)^{m+1}\,(b\,\mathrm{Sin}[e+f\,x])^n}{d\,(m+1)}-\frac{b\,f\,n\,(c+d\,x)^{m+2}\,\mathrm{Cos}[e+f\,x]\,(b\,\mathrm{Sin}[e+f\,x])^{n-1}}{d^2\,(m+1)\,(m+2)}-$$

$$\frac{f^2\,n^2}{d^2\,(m+1)\,(m+2)}\int (c+d\,x)^{m+2}\,(b\,\mathrm{Sin}[e+f\,x])^n\,dx+\frac{b^2\,f^2\,n\,(n-1)}{d^2\,(m+1)\,(m+2)}\int (c+d\,x)^{m+2}\,(b\,\mathrm{Sin}[e+f\,x])^{n-2}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_*(b_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  (c+d*x)^(m+1)*(b*Sin[e+f*x])^n/(d*(m+1)) -
  b*f*n*(c+d*x)^(m+2)*Cos[e+f*x]*(b*Sin[e+f*x])^(n-1)/(d^2*(m+1)*(m+2)) -
  f^2*n^2/(d^2*(m+1)*(m+2))*Int[(c+d*x)^(m+2)*(b*Sin[e+f*x])^n,x] +
  b^2*f^2*n*(n-1)/(d^2*(m+1)*(m+2))*Int[(c+d*x)^(m+2)*(b*Sin[e+f*x])^(n-2),x] /;
FreeQ[{b,c,d,e,f},x] && GtQ[n,1] && LtQ[m,-2]
```

2. $\int (c + d\,x)^m\,(b\,\mathrm{Sin}[e + f\,x])^n\,dx$ when $n < -1$

1: $\int (c + d\,x)\,(b\,\mathrm{Sin}[e + f\,x])^n\,dx$ when $n < -1 \wedge n \neq -2$

Reference: G&R 2.643.1 with $m \to 1$

Reference: G&R 2.643.2 with $m \to 1$

Rule: If $n < -1 \wedge n \neq -2$, then

$$\int (c + d\,x)\,(b\,\mathrm{Sin}[e + f\,x])^n\,dx \longrightarrow$$

$$\frac{(c + d\,x)\,\mathrm{Cos}[e + f\,x]\,(b\,\mathrm{Sin}[e + f\,x])^{n+1}}{b\,f\,(n + 1)} - \frac{d\,(b\,\mathrm{Sin}[e + f\,x])^{n+2}}{b^2\,f^2\,(n + 1)\,(n + 2)} + \frac{n + 2}{b^2\,(n + 1)} \int (c + d\,x)\,(b\,\mathrm{Sin}[e + f\,x])^{n+2}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)*(b_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  (c+d*x)*Cos[e+f*x]*(b*Sin[e+f*x])^(n+1)/(b*f*(n+1)) -
  d*(b*Sin[e+f*x])^(n+2)/(b^2*f^2*(n+1)*(n+2)) +
  (n+2)/(b^2*(n+1))*Int[(c+d*x)*(b*Sin[e+f*x])^(n+2),x] /;
FreeQ[{b,c,d,e,f},x] && LtQ[n,-1] && NeQ[n,-2]
```

2: $\int (c+d\,x)^m\,(b\,\text{Sin}[e+f\,x])^n\,dx$ when $n<-1\land n\neq-2\land m>1$

Reference: G&R 2.643.1

Reference: G&R 2.643.2

Rule: If $n<-1\land n\neq-2\land m>1$, then

$$\int (c+d\,x)^m\,(b\,\text{Sin}[e+f\,x])^n\,dx\longrightarrow$$
$$\frac{(c+d\,x)^m\,\text{Cos}[e+f\,x]\,(b\,\text{Sin}[e+f\,x])^{n+1}}{b\,f\,(n+1)}-\frac{d\,m\,(c+d\,x)^{m-1}\,(b\,\text{Sin}[e+f\,x])^{n+2}}{b^2\,f^2\,(n+1)\,(n+2)}+$$
$$\frac{n+2}{b^2\,(n+1)}\int (c+d\,x)^m\,(b\,\text{Sin}[e+f\,x])^{n+2}\,dx+\frac{d^2\,m\,(m-1)}{b^2\,f^2\,(n+1)\,(n+2)}\int (c+d\,x)^{m-2}\,(b\,\text{Sin}[e+f\,x])^{n+2}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(b_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  (c+d*x)^m*Cos[e+f*x]*(b*Sin[e+f*x])^(n+1)/(b*f*(n+1)) -
  d*m*(c+d*x)^(m-1)*(b*Sin[e+f*x])^(n+2)/(b^2*f^2*(n+1)*(n+2)) +
  (n+2)/(b^2*(n+1))*Int[(c+d*x)^m*(b*Sin[e+f*x])^(n+2),x] +
  d^2*m*(m-1)/(b^2*f^2*(n+1)*(n+2))*Int[(c+d*x)^(m-2)*(b*Sin[e+f*x])^(n+2),x] /;
FreeQ[{b,c,d,e,f},x] && LtQ[n,-1] && NeQ[n,-2] && GtQ[m,1]
```

2: $\int (c+d\,x)^m\,(a+b\,\text{Sin}[e+f\,x])^n\,dx$ when $n\in\mathbb{Z}^+\land\left(n==1\lor m\in\mathbb{Z}^+\lor a^2-b^2\neq0\right)$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+\land\left(n==1\lor m\in\mathbb{Z}^+\lor a^2-b^2\neq0\right)$, then

$$\int (c+d\,x)^m\,(a+b\,\text{Sin}[e+f\,x])^n\,dx\longrightarrow\int (c+d\,x)^m\,\text{ExpandIntegrand}[(a+b\,\text{Sin}[e+f\,x])^n,\,x]\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_+b_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(c+d*x)^m,(a+b*Sin[e+f*x])^n,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[n,0] && (EqQ[n,1] || IGtQ[m,0] || NeQ[a^2-b^2,0])
```

3. $\int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0\ \land\ 2\,n\in\mathbb{Z}\ \land\ (n>0\ \lor\ m\in\mathbb{Z}^+)$

1: $\int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0\ \land\ n\in\mathbb{Z}\ \land\ (n>0\ \lor\ m\in\mathbb{Z}^+)$

Derivation: Algebraic simplification

Basis: If $a^2-b^2==0$, **then** $a+b\,\mathrm{Sin}[e+f\,x]==2\,a\,\mathrm{Sin}\left[\frac{1}{2}\left(e+\frac{\pi\,a}{2\,b}\right)+\frac{f\,x}{2}\right]^2$

Rule: If $a^2-b^2==0\ \land\ n\in\mathbb{Z}\ \land\ (n>0\ \lor\ m\in\mathbb{Z}^+)$, **then**

$$\int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^n\,dx\ \longrightarrow\ (2\,a)^n\int (c+d\,x)^m\,\mathrm{Sin}\left[\frac{1}{2}\left(e+\frac{\pi\,a}{2\,b}\right)+\frac{f\,x}{2}\right]^{2\,n}dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_+b_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  (2*a)^n*Int[(c+d*x)^m*Sin[1/2*(e+Pi*a/(2*b))+f*x/2]^(2*n),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[a^2-b^2,0] && IntegerQ[n] && (GtQ[n,0] || IGtQ[m,0])
```

2: $\int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0\ \bigwedge\ n+\frac{1}{2}\in\mathbb{Z}\ \bigwedge\ (n>0\ \lor\ m\in\mathbb{Z}^+)$

Derivation: Piecewise constant extraction

Basis: If $a^2-b^2==0$, **then** $\partial_x\frac{(a+b\,\mathrm{Sin}[e+f\,x])^n}{\mathrm{Sin}\left[\frac{1}{2}\left(e+\frac{\pi\,a}{2\,b}\right)+\frac{f\,x}{2}\right]^{2\,n}}==0$

Rule: If $a^2-b^2==0\ \bigwedge\ n+\frac{1}{2}\in\mathbb{Z}\ \bigwedge\ (n>0\ \lor\ m\in\mathbb{Z}^+)$, **then**

$$\int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^n\,dx\ \longrightarrow\ \frac{(2\,a)^{\mathrm{IntPart}[n]}\,(a+b\,\mathrm{Sin}[e+f\,x])^{\mathrm{FracPart}[n]}}{\mathrm{Sin}\left[\frac{e}{2}+\frac{a\,\pi}{4\,b}+\frac{f\,x}{2}\right]^{2\,\mathrm{FracPart}[n]}}\int (c+d\,x)^m\,\mathrm{Sin}\left[\frac{e}{2}+\frac{a\,\pi}{4\,b}+\frac{f\,x}{2}\right]^{2\,n}dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_+b_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  (2*a)^IntPart[n]*(a+b*Sin[e+f*x])^FracPart[n]/Sin[e/2+a*Pi/(4*b)+f*x/2]^(2*FracPart[n])*
    Int[(c+d*x)^m*Sin[e/2+a*Pi/(4*b)+f*x/2]^(2*n),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[a^2-b^2,0] && IntegerQ[n+1/2] && (GtQ[n,0] || IGtQ[m,0])
```

x: $\int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\land\,n\in\mathbb{Z}\,\land\,(n>0\,\lor\,m\in\mathbb{Z}^+)$

Derivation: Algebraic simplification

- **Basis: If** $a^2-b^2==0$, **then** $a+b\,\mathrm{Sin}[z]==2\,a\,\mathrm{Cos}\left[-\frac{\pi\,a}{4\,b}+\frac{z}{2}\right]^2$

- **Rule: If** $a^2-b^2==0\,\land\,n\in\mathbb{Z}\,\land\,(n>0\,\lor\,m\in\mathbb{Z}^+)$, **then**

$$\int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^n\,dx\,\longrightarrow\,(2\,a)^n\int (c+d\,x)^m\,\mathrm{Cos}\left[\frac{1}{2}\left(e-\frac{\pi\,a}{2\,b}\right)+\frac{f\,x}{2}\right]^{2\,n}dx$$

Program code:

```
(* Int[(c_.+d_.*x_)^m_.*(a_+b_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  (2*a)^n*Int[(c+d*x)^m*Cos[1/2*(e-Pi*a/(2*b))+f*x/2]^(2*n),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[a^2-b^2,0] && IntegerQ[n] && (GtQ[n,0] || IGtQ[m,0]) *)
```

x: $\int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\land\,n+\frac{1}{2}\in\mathbb{Z}\,\land\,(n>0\,\lor\,m\in\mathbb{Z}^+)$

Derivation: Piecewise constant extraction

- **Basis: If** $a^2-b^2==0$, **then** $\partial_x\frac{(a+b\,\mathrm{Sin}[e+f\,x])^n}{\mathrm{Cos}\left[\frac{1}{2}\left(e-\frac{\pi\,a}{2\,b}\right)+\frac{f\,x}{2}\right]^{2\,n}}==0$

- **Rule: If** $a^2-b^2==0\,\land\,n+\frac{1}{2}\in\mathbb{Z}\,\land\,(n>0\,\lor\,m\in\mathbb{Z}^+)$, **then**

$$\int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^n\,dx\,\longrightarrow\,\frac{(2\,a)^{\mathrm{IntPart}[n]}\,(a+b\,\mathrm{Sin}[e+f\,x])^{\mathrm{FracPart}[n]}}{\mathrm{Cos}\left[\frac{1}{2}\left(e-\frac{\pi\,a}{2\,b}\right)+\frac{f\,x}{2}\right]^{2\,\mathrm{FracPart}[n]}}\int (c+d\,x)^m\,\mathrm{Cos}\left[\frac{1}{2}\left(e-\frac{\pi\,a}{2\,b}\right)+\frac{f\,x}{2}\right]^{2\,n}dx$$

Program code:

```
(* Int[(c_.+d_.*x_)^m_.*(a_+b_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  (2*a)^IntPart[n]*(a+b*Sin[e+f*x])^FracPart[n]/Cos[1/2*(e-Pi*a/(2*b))+f*x/2]^(2*FracPart[n])*
    Int[(c+d*x)^m*Cos[1/2*(e-Pi*a/(2*b))+f*x/2]^(2*n),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[a^2-b^2,0] && IntegerQ[n+1/2] && (GtQ[n,0] || IGtQ[m,0]) *)
```

4. $\int (c+d\,x)^m\,(a+b\,\text{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\,\wedge\,n\in\mathbb{Z}^-\,\wedge\,m\in\mathbb{Z}^+$

1: $\int \frac{(c+d\,x)^m}{a+b\,\text{Sin}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0\,\wedge\,m\in\mathbb{Z}^+$

Derivation: Algebraic expansion

- **Basis:** $\frac{1}{a+b\,\text{Sin}[z]}==\frac{2\,e^{i\,z}}{i\,b+2\,a\,e^{i\,z}-i\,b\,e^{2\,i\,z}}==\frac{2\,e^{-i\,z}}{-i\,b+2\,a\,e^{-i\,z}+i\,b\,e^{-2\,i\,z}}$
- **Basis:** $\frac{1}{a+b\,\text{Cos}[z]}==\frac{2\,e^{i\,z}}{b+2\,a\,e^{i\,z}+b\,e^{2\,i\,z}}$

Rule: If $a^2-b^2\neq 0\,\wedge\,m\in\mathbb{Z}^+$**, then**

$$\int \frac{(c+d\,x)^m}{a+b\,\text{Sin}[e+f\,x]}\,dx\;\longrightarrow\;-2\,i\int \frac{(c+d\,x)^m\,e^{i\,(e+f\,x)}}{b-2\,i\,a\,e^{i\,(e+f\,x)}-b\,e^{2\,i\,(e+f\,x)}}\,dx$$

$$\int \frac{(c+d\,x)^m}{a+b\,\text{Cos}[e+f\,x]}\,dx\;\longrightarrow\;2\int \frac{(c+d\,x)^m\,e^{i\,(e+f\,x)}}{b+2\,a\,e^{i\,(e+f\,x)}+b\,e^{2\,i\,(e+f\,x)}}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_./(a_+b_.*sin[e_.+k_.*Pi+f_.*Complex[0,fz_]*x_]),x_Symbol] :=
  2*Int[(c+d*x)^m*E^(-I*Pi*(k-1/2))*E^(-I*e+f*fz*x)/(b+2*a*E^(-I*Pi*(k-1/2))*E^(-I*e+f*fz*x)-b*E^(-2*I*k*Pi)*E^(2*(-I*e+f*fz*x))),x
FreeQ[{a,b,c,d,e,f,fz},x] && IntegerQ[2*k] && NeQ[a^2-b^2,0] && IGtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_./(a_+b_.*sin[e_.+k_.*Pi+f_.*x_]),x_Symbol] :=
  2*Int[(c+d*x)^m*E^(I*Pi*(k-1/2))*E^(I*(e+f*x))/(b+2*a*E^(I*Pi*(k-1/2))*E^(I*(e+f*x))-b*E^(2*I*k*Pi)*E^(2*I*(e+f*x))),x] /;
FreeQ[{a,b,c,d,e,f},x] && IntegerQ[2*k] && NeQ[a^2-b^2,0] && IGtQ[m,0]
```

```
(* Int[(c_.+d_.*x_)^m_./(a_+b_.*sin[e_.+f_.*Complex[0,fz_]*x_]),x_Symbol] :=
  2*I*Int[(c+d*x)^m*E^(-I*e+f*fz*x)/(b+2*I*a*E^(-I*e+f*fz*x)-b*E^(2*(-I*e+f*fz*x))),x] /;
FreeQ[{a,b,c,d,e,f,fz},x] && NeQ[a^2-b^2,0] && IGtQ[m,0] *)
```

```
(* Int[(c_.+d_.*x_)^m_./(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -2*I*Int[(c+d*x)^m*E^(I*(e+f*x))/(b-2*I*a*E^(I*(e+f*x))-b*E^(2*I*(e+f*x))),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[a^2-b^2,0] && IGtQ[m,0] *)
```

```
Int[(c_.+d_.*x_)^m_./(a_+b_.*sin[e_.+f_.*Complex[0,fz_]*x_]),x_Symbol] :=
  2*Int[(c+d*x)^m*E^(-I*e+f*fz*x)/(-I*b+2*a*E^(-I*e+f*fz*x)+I*b*E^(2*(-I*e+f*fz*x))),x] /;
FreeQ[{a,b,c,d,e,f,fz},x] && NeQ[a^2-b^2,0] && IGtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_./(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  2*Int[(c+d*x)^m*E^(I*(e+f*x))/(I*b+2*a*E^(I*(e+f*x))-I*b*E^(2*I*(e+f*x))),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[a^2-b^2,0] && IGtQ[m,0]
```

2: $\int \frac{(c+d\,x)^m}{(a+b\,\mathrm{Sin}[e+f\,x])^2}\,dx$ when $a^2-b^2\neq 0\ \wedge\ m\in\mathbb{Z}^+$

Rule: If $a^2-b^2\neq 0\ \wedge\ m\in\mathbb{Z}^+$, then

$$\int \frac{(c+d\,x)^m}{(a+b\,\mathrm{Sin}[e+f\,x])^2}\,dx\ \longrightarrow\ \frac{b\,(c+d\,x)^m\,\mathrm{Cos}[e+f\,x]}{f\left(a^2-b^2\right)(a+b\,\mathrm{Sin}[e+f\,x])}+\frac{a}{a^2-b^2}\int \frac{(c+d\,x)^m}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx-\frac{b\,d\,m}{f\left(a^2-b^2\right)}\int \frac{(c+d\,x)^{m-1}\,\mathrm{Cos}[e+f\,x]}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_./(a_+b_.*sin[e_.+f_.*x_])^2,x_Symbol] :=
  b*(c+d*x)^m*Cos[e+f*x]/(f*(a^2-b^2)*(a+b*Sin[e+f*x])) +
  a/(a^2-b^2)*Int[(c+d*x)^m/(a+b*Sin[e+f*x]),x] -
  b*d*m/(f*(a^2-b^2))*Int[(c+d*x)^(m-1)*Cos[e+f*x]/(a+b*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[a^2-b^2,0] && IGtQ[m,0]
```

3: $\int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\ \wedge\ n+2\in\mathbb{Z}^-\ \wedge\ m\in\mathbb{Z}^+$

Rule: If $a^2-b^2\neq 0\ \wedge\ n+2\in\mathbb{Z}^-\ \wedge\ m\in\mathbb{Z}^+$, then

$$\int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^n\,dx\ \longrightarrow$$

$$-\frac{b\,(c+d\,x)^m\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{n+1}}{f\,(n+1)\left(a^2-b^2\right)}+\frac{a}{a^2-b^2}\int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^{n+1}\,dx+$$

$$\frac{b\,d\,m}{f\,(n+1)\left(a^2-b^2\right)}\int (c+d\,x)^{m-1}\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{n+1}\,dx-\frac{b\,(n+2)}{(n+1)\left(a^2-b^2\right)}\int (c+d\,x)^m\,\mathrm{Sin}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{n+1}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_+b_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -b*(c+d*x)^m*Cos[e+f*x]*(a+b*Sin[e+f*x])^(n+1)/(f*(n+1)*(a^2-b^2)) +
  a/(a^2-b^2)*Int[(c+d*x)^m*(a+b*Sin[e+f*x])^(n+1),x] +
  b*d*m/(f*(n+1)*(a^2-b^2))*Int[(c+d*x)^(m-1)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(n+1),x] -
  b*(n+2)/((n+1)*(a^2-b^2))*Int[(c+d*x)^m*Sin[e+f*x]*(a+b*Sin[e+f*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[a^2-b^2,0] && ILtQ[n,-2] && IGtQ[m,0]
```

X: $\int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^n\,dx$

- **Rule:**

$$\int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^n\,dx \longrightarrow \int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^n\,dx$$

- **Program code:**

```
Int[(c_.+d_.*x_)^m_.*(a_.+b_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(c+d*x)^m*(a+b*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n},x]
```

N: $\int u^m\,(a+b\,\mathrm{Sin}[v])^n\,dx$ when $u == c+d\,x \wedge v == e+f\,x$

- **Derivation: Algebraic normalization**
- **Rule: If** $u == c+d\,x \wedge v == e+f\,x$**, then**

$$\int u^m\,(a+b\,\mathrm{Sin}[v])^n\,dx \longrightarrow \int (c+d\,x)^m\,(a+b\,\mathrm{Sin}[e+f\,x])^n\,dx$$

- **Program code:**

```
Int[u_^m_.*(a_.+b_.*Sin[v_])^n_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*(a+b*Sin[ExpandToSum[v,x]])^n,x] /;
FreeQ[{a,b,m,n},x] && LinearQ[{u,v},x] && Not[LinearMatchQ[{u,v},x]]
```

```
Int[u_^m_.*(a_.+b_.*Cos[v_])^n_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*(a+b*Cos[ExpandToSum[v,x]])^n,x] /;
FreeQ[{a,b,m,n},x] && LinearQ[{u,v},x] && Not[LinearMatchQ[{u,v},x]]
```