Rules for integrands of the form $(f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p$

0. $\int (f\,x)^m\,(e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$

   1. $\int (f\,x)^m\,(e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $m\in\mathbb{Z}\,\lor\,f>0$

      1: $\int (f\,x)^m\,(e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $(m\in\mathbb{Z}\,\lor\,f>0)\,\land\,\frac{m+1}{n}\in\mathbb{Z}$

- **Derivation: Integration by substitution**
- **Basis: If** $\frac{m+1}{n}\in\mathbb{Z}$, **then** $x^m\,(e\,x^n)^q = \frac{1}{e^{\frac{m+1}{n}-1}}\,x^{n-1}\,(e\,x^n)^{q+\frac{m+1}{n}-1}$
- **Basis:** $x^{n-1}\,F[x^n] = \frac{1}{n}\,\text{Subst}[F[x],\,x,\,x^n]\,\partial_x x^n$
- **Rule 1.2.3.4.0.1.1: If** $(m\in\mathbb{Z}\,\lor\,f>0)\,\land\,\frac{m+1}{n}\in\mathbb{Z}$, **then**

$$\int (f\,x)^m\,(e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\to\;\frac{f^m}{n\,e^{\frac{m+1}{n}-1}}\,\text{Subst}\left[\int (e\,x)^{q+\frac{m+1}{n}-1}\,\left(a+b\,x+c\,x^2\right)^p\,dx,\;x,\;x^n\right]$$

- **Program code:**

```
Int[(f_.*x_)^m_.*(e_.*x_^n_)^q_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  f^m/(n*e^((m+1)/n-1))*Subst[Int[(e*x)^(q+(m+1)/n-1)*(a+b*x+c*x^2)^p,x],x,x^n] /;
FreeQ[{a,b,c,e,f,m,n,p,q},x] && EqQ[n2,2*n] && (IntegerQ[m] || GtQ[f,0]) && IntegerQ[Simplify[(m+1)/n]]
```

```
Int[(f_.*x_)^m_.*(e_.*x_^n_)^q_*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  f^m/(n*e^((m+1)/n-1))*Subst[Int[(e*x)^(q+(m+1)/n-1)*(a+c*x^2)^p,x],x,x^n] /;
FreeQ[{a,c,e,f,m,n,p,q},x] && EqQ[n2,2*n] && (IntegerQ[m] || GtQ[f,0]) && IntegerQ[Simplify[(m+1)/n]]
```

2: $\int (f\,x)^m\,(e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $(m\in\mathbb{Z}\,\lor\,f>0)\,\bigwedge\,\frac{m+1}{n}\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x\frac{(e\,x^n)^q}{x^{n\,q}}=0$
- **Rule 1.2.3.4.0.1.2: If** $(m\in\mathbb{Z}\,\lor\,f>0)\,\bigwedge\,\frac{m+1}{n}\notin\mathbb{Z}$, **then**

$$\int (f\,x)^m\,(e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow\,\frac{f^m\,e^{\text{IntPart}[q]}\,(e\,x^n)^{\text{FracPart}[q]}}{x^{n\,\text{FracPart}[q]}}\int x^{m+n\,q}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(e_.*x_^n_)^q_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  f^m*e^IntPart[q]*(e*x^n)^FracPart[q]/x^(n*FracPart[q])*Int[x^(m+n*q)*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,e,f,m,n,p,q},x] && EqQ[n2,2*n] && (IntegerQ[m] || GtQ[f,0]) && Not[IntegerQ[Simplify[(m+1)/n]]]
```

```
Int[(f_.*x_)^m_.*(e_.*x_^n_)^q_*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  f^m*e^IntPart[q]*(e*x^n)^FracPart[q]/x^(n*FracPart[q])*Int[x^(m+n*q)*(a+c*x^(2*n))^p,x] /;
FreeQ[{a,c,e,f,m,n,p,q},x] && EqQ[n2,2*n] && (IntegerQ[m] || GtQ[f,0]) && Not[IntegerQ[Simplify[(m+1)/n]]]
```

2: $\int (f\,x)^m\,(e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x\frac{(f\,x)^m}{x^m}=0$

Rule 1.2.3.4.0.2: If $m\notin\mathbb{Z}$, **then**

$$\int (f\,x)^m\,(e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow\,\frac{f^{\text{IntPart}[m]}\,(f\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,(e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[(f_*x_)^m_.*(e_.*x_^n_)^q_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  f^IntPart[m]*(f*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(e*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,e,f,m,n,p,q},x] && EqQ[n2,2*n] && Not[IntegerQ[m]]
```

```
Int[(f_*x_)^m_.*(e_.*x_^n_)^q_*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  f^IntPart[m]*(f*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(e*x^n)^q*(a+c*x^(2*n))^p,x] /;
FreeQ[{a,c,e,f,m,n,p,q},x] && EqQ[n2,2*n] && Not[IntegerQ[m]]
```

1: $\int x^m (d+e\,x^n)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $m-n+1=0$

Derivation: Integration by substitution

Basis: $x^{n-1}\,F[x^n] = \frac{1}{n}\,\text{Subst}[F[x],\,x,\,x^n]\,\partial_x x^n$

Rule 1.2.3.4.1: If $m-n+1=0$, **then**

$$\int x^m (d+e\,x^n)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx \longrightarrow \frac{1}{n}\,\text{Subst}\left[\int (d+e\,x)^q \left(a+b\,x+c\,x^2\right)^p dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  1/n*Subst[Int[(d+e*x)^q*(a+b*x+c*x^2)^p,x],x,x^n] /;
FreeQ[{a,b,c,d,e,m,n,p,q},x] && EqQ[n2,2*n] && EqQ[Simplify[m-n+1],0]
```

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  1/n*Subst[Int[(d+e*x)^q*(a+c*x^2)^p,x],x,x^n] /;
FreeQ[{a,c,d,e,m,n,p,q},x] && EqQ[n2,2*n] && EqQ[Simplify[m-n+1],0]
```

2: $\int x^m (d+e\,x^n)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $(p\mid q)\in\mathbb{Z}\wedge n<0$

Derivation: Algebraic expansion

Basis: If $(p\mid q)\in\mathbb{Z}$, **then** $(d+e\,x^n)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p = x^{n\,(2\,p+q)}\,(e+d\,x^{-n})^q \left(c+b\,x^{-n}+a\,x^{-2\,n}\right)^p$

Rule 1.2.3.4.2: If $(p\mid q)\in\mathbb{Z}\wedge n<0$, **then**

$$\int x^m (d+e\,x^n)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx \longrightarrow \int x^{m+n\,(2\,p+q)}\,(e+d\,x^{-n})^q \left(c+b\,x^{-n}+a\,x^{-2\,n}\right)^p dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[x^(m+n*(2*p+q))*(e+d*x^(-n))^q*(c+b*x^(-n)+a*x^(-2*n))^p,x] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[n2,2*n] && IntegersQ[p,q] && NegQ[n]
```

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[x^(m+n*(2*p+q))*(e+d*x^(-n))^q*(c+a*x^(-2*n))^p,x] /;
FreeQ[{a,c,d,e,m,n},x] && EqQ[n2,2*n] && IntegersQ[p,q] && NegQ[n]
```

3. $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c=0\,\wedge\,p\notin\mathbb{Z}$

1: $\int x^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c=0\,\bigwedge\,p\notin\mathbb{Z}\,\bigwedge\,\left(m\,\middle|\,n\,\middle|\,\frac{m+1}{n}\right)\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n}\in\mathbb{Z}$, **then** $x^m\,F[x^n]=\frac{1}{n}\,\text{Subst}\left[x^{\frac{m+1}{n}-1}\,F[x],\,x,\,x^n\right]\,\partial_x x^n$

Note: If this substitution rule is applied when $m\in\mathbb{Z}^-$, **expressions of the form** $\text{Log}[x^n]$ **rather than** $\text{Log}[x]$ **may appear in the antiderivative.**

Rule 1.2.3.4.3.1: If $b^2-4\,a\,c=0\,\bigwedge\,p\notin\mathbb{Z}\,\bigwedge\,\left(m\,\middle|\,n\,\middle|\,\frac{m+1}{n}\right)\in\mathbb{Z}^+$, **then**

$$\int x^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\longrightarrow\;\frac{1}{n}\,\text{Subst}\left[\int x^{\frac{m+1}{n}-1}\,(d+e\,x)^q\,\left(a+b\,x+c\,x^2\right)^p\,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  1/n*Subst[Int[x^((m+1)/n-1)*(d+e*x)^q*(a+b*x+c*x^2)^p,x],x,x^n] /;
FreeQ[{a,b,c,d,e,p,q},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[(m+1)/n,0]
```

2: $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0\,\wedge\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis: If** $b^2-4\,a\,c==0$**, then** $\partial_x\frac{(a+b\,x^n+c\,x^{2\,n})^p}{\left(\frac{b}{2}+c\,x^n\right)^{2\,p}}==0$

- **Basis: If** $b^2-4\,a\,c==0$**, then** $\frac{(a+b\,x^n+c\,x^{2\,n})^p}{\left(\frac{b}{2}+c\,x^n\right)^{2\,p}}==\frac{(a+b\,x^n+c\,x^{2\,n})^{\text{FracPart}[p]}}{c^{\text{IntPart}[p]}\,\left(\frac{b}{2}+c\,x^n\right)^{2\,\text{FracPart}[p]}}$

Rule 1.2.3.4.3.2: If $b^2-4\,a\,c==0\,\wedge\,p\notin\mathbb{Z}$**, then**

$$\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow\,\frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^{\text{FracPart}[p]}}{c^{\text{IntPart}[p]}\,\left(\frac{b}{2}+c\,x^n\right)^{2\,\text{FracPart}[p]}}\int (f\,x)^m\,(d+e\,x^n)^q\,\left(\frac{b}{2}+c\,x^n\right)^{2\,p}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  (a+b*x^n+c*x^(2*n))^FracPart[p]/(c^IntPart[p]*(b/2+c*x^n)^(2*FracPart[p]))*
    Int[(f*x)^m*(d+e*x^n)^q*(b/2+c*x^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

4. $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}$

1: $\int x^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}$

Derivation: Integration by substitution

- **Basis: If** $\frac{m+1}{n}\in\mathbb{Z}$**, then** $x^m\,F[x^n] = \frac{1}{n}\,\text{Subst}\left[x^{\frac{m+1}{n}-1}\,F[x],\,x,\,x^n\right]\,\partial_x x^n$
- **Note: If** $n\in\mathbb{Z}\bigwedge\frac{m+1}{n}\in\mathbb{Z}$**, then** $m\in\mathbb{Z}$**, and** $(f\,x)^m$ **automatically evaluates to** $f^m\,x^m$**.**
- **Rule 1.2.3.4.4.1: If** $\frac{m+1}{n}\in\mathbb{Z}$**, then**

$$\int x^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\longrightarrow\;\frac{1}{n}\,\text{Subst}\left[\int x^{\frac{m+1}{n}-1}\,(d+e\,x)^q\,\left(a+b\,x+c\,x^2\right)^p\,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(d+e*x)^q*(a+b*x+c*x^2)^p,x],x,x^n] /;
FreeQ[{a,b,c,d,e,m,n,p,q},x] && EqQ[n2,2*n] && IntegerQ[Simplify[(m+1)/n]]
```

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(d+e*x)^q*(a+c*x^2)^p,x],x,x^n] /;
FreeQ[{a,c,d,e,m,n,p,q},x] && EqQ[n2,2*n] && IntegerQ[Simplify[(m+1)/n]]
```

2: $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

- Basis: $\partial_x \frac{(f\,x)^m}{x^m} == 0$
- Basis: $\frac{(f\,x)^m}{x^m} == \frac{f^{\text{IntPart}[m]}\,(f\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$
- Rule 1.2.3.4.4.2: If $\frac{m+1}{n}\in\mathbb{Z}$, then

$$\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx \longrightarrow \frac{f^{\text{IntPart}[m]}\,(f\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[(f_*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  f^IntPart[m]*(f*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(d+e*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q},x] && EqQ[n2,2*n] && IntegerQ[Simplify[(m+1)/n]]
```

```
Int[(f_*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  f^IntPart[m]*(f*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(d+e*x^n)^q*(a+c*x^(2*n))^p,x] /;
FreeQ[{a,c,d,e,f,m,n,p,q},x] && EqQ[n2,2*n] && IntegerQ[Simplify[(m+1)/n]]
```

5. $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0$

1: $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $c\,d^2-b\,d\,e+a\,e^2==0$, **then** $a+b\,z+c\,z^2==(d+e\,z)\left(\frac{a}{d}+\frac{c\,z}{e}\right)$

Rule 1.2.3.4.5.1: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\in\mathbb{Z}$, **then**

$$\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx \longrightarrow \int (f\,x)^m\,(d+e\,x^n)^{q+p}\left(\frac{a}{d}+\frac{c\,x^n}{e}\right)^p\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_)^p_.,x_Symbol] :=
  Int[(f*x)^m*(d+e*x^n)^(q+p)*(a/d+c/e*x^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[p]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_)^p_.,x_Symbol] :=
  Int[(f*x)^m*(d+e*x^n)^(q+p)*(a/d+c/e*x^n)^p,x] /;
FreeQ[{a,c,d,e,f,q,m,n,q},x] && EqQ[n2,2*n] && EqQ[c*d^2+a*e^2,0] && IntegerQ[p]
```

2: $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2==0\ \land\ p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis: If** $c\,d^2-b\,d\,e+a\,e^2==0$**, then** $\partial_x \frac{(a+b\,x^n+c\,x^{2\,n})^p}{(d+e\,x^n)^p\,\left(\frac{a}{d}+\frac{c\,x^n}{e}\right)^p}==0$
- **Basis: If** $c\,d^2-b\,d\,e+a\,e^2==0$**, then** $\frac{(a+b\,x^n+c\,x^{2\,n})^p}{(d+e\,x^n)^p\,\left(\frac{a}{d}+\frac{c\,x^n}{e}\right)^p}==\frac{(a+b\,x^n+c\,x^{2\,n})^{\text{FracPart}[p]}}{(d+e\,x^n)^{\text{FracPart}[p]}\,\left(\frac{a}{d}+\frac{c\,x^n}{e}\right)^{\text{FracPart}[p]}}$

Rule 1.2.3.4.5.2: If $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2==0\ \land\ p\notin\mathbb{Z}$**, then**

$$\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\ \longrightarrow\ \frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^{\text{FracPart}[p]}}{(d+e\,x^n)^{\text{FracPart}[p]}\,\left(\frac{a}{d}+\frac{c\,x^n}{e}\right)^{\text{FracPart}[p]}}\int (f\,x)^m\,(d+e\,x^n)^{q+p}\,\left(\frac{a}{d}+\frac{c\,x^n}{e}\right)^p\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  (a+b*x^n+c*x^(2*n))^FracPart[p]/((d+e*x^n)^FracPart[p]*(a/d+(c*x^n)/e)^FracPart[p])*
    Int[(f*x)^m*(d+e*x^n)^(q+p)*(a/d+c/e*x^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_*(a_+c_.*x_^n2_)^p_,x_Symbol] :=
  (a+c*x^(2*n))^FracPart[p]/((d+e*x^n)^FracPart[p]*(a/d+(c*x^n)/e)^FracPart[p])*Int[(f*x)^m*(d+e*x^n)^(q+p)*(a/d+c/e*x^n)^p,x] /
FreeQ[{a,c,d,e,f,m,n,p,q},x] && EqQ[n2,2*n] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]]
```

6. $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}$

1. $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+$

1. $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,(n\mid p)\in\mathbb{Z}^+$

1. $\int x^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,(n\mid p)\in\mathbb{Z}^+\,\wedge\,(m\mid q)\in\mathbb{Z}\,\wedge\,q<-1$

1: $\int x^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,(n\mid p)\in\mathbb{Z}^+\,\wedge\,(m\mid q)\in\mathbb{Z}\,\wedge\,q<-1\,\wedge\,m>0$

Derivation: Algebraic expansion and binomial recurrence 2b

Note: If $(n\mid p)\in\mathbb{Z}^+\,\wedge\,(m\mid q)\in\mathbb{Z}\,\wedge\,q<0$**, then** $\frac{(-d)^{(m-\text{Mod}[m,n])/n}}{e^{2\,p+(m-\text{Mod}[m,n])/n}}\sum_{k=0}^{2\,p}(-d)^k\,e^{2\,p-k}\,P_{2\,p}[x^n,\,k]$ **is the coefficient of the** $x^{\text{Mod}[m,n]}\,(d+e\,x^n)^q$ **term of the partial fraction expansion of** $x^m\,P_{2\,p}[x^n]\,(d+e\,x^n)^q$**.**

Note: If $(n\mid p)\in\mathbb{Z}^+\,\wedge\,(m\mid q)\in\mathbb{Z}\,\wedge\,q<-1\,\wedge\,m>0$**, then**
$n\,e^{2\,p+(m-\text{Mod}[m,n])/n}\,(q+1)\,x^{m-\text{Mod}[m,n]}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p-(-d)^{(m-\text{Mod}[m,n])/n-1}\,\left(c\,d^2-b\,d\,e+a\,e^2\right)^p\,(d\,(\text{Mod}[m,\,n]+1)+e\,(\text{Mod}[m,\,n]+n\,(q+1)+1)\,x^n)$
will be divisible by $a+b\,x^n$**.**

Note: In the resulting integrand the degree of the polynomial in x^n **is at most** $q-1$ **.**

Rule 1.2.3.4.6.1.1.1.1: If $b^2-4\,a\,c\neq 0\,\wedge\,(n\mid p)\in\mathbb{Z}^+\,\wedge\,(m\mid q)\in\mathbb{Z}\,\wedge\,q<-1\,\wedge\,m>0$**, then**

$$\int x^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow$$

$$\frac{(-d)^{(m-\text{Mod}[m,n])/n}}{e^{2\,p+(m-\text{Mod}[m,n])/n}}\left(c\,d^2-b\,d\,e+a\,e^2\right)^p\int x^{\text{Mod}[m,n]}\,(d+e\,x^n)^q\,dx+$$
$$\frac{1}{e^{2\,p+(m-\text{Mod}[m,n])/n}}\int x^{\text{Mod}[m,n]}\,(d+e\,x^n)^q\,\left(e^{2\,p+(m-\text{Mod}[m,n])/n}\,x^{m-\text{Mod}[m,n]}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p-(-d)^{(m-\text{Mod}[m,n])/n}\,\left(c\,d^2-b\,d\,e+a\,e^2\right)^p\right)dx\,\longrightarrow$$

$$\frac{(-d)^{(m-\text{Mod}[m,n])/n-1}\,\left(c\,d^2-b\,d\,e+a\,e^2\right)^p\,x^{\text{Mod}[m,n]+1}\,(d+e\,x^n)^{q+1}}{n\,e^{2\,p+(m-\text{Mod}[m,n])/n}\,(q+1)}+$$
$$\frac{1}{n\,e^{2\,p+(m-\text{Mod}[m,n])/n}\,(q+1)}\int x^{\text{Mod}[m,n]}\,(d+e\,x^n)^{q+1}\,\cdot$$
$$\left(\frac{1}{d+e\,x^n}\left(n\,e^{2\,p+(m-\text{Mod}[m,n])/n}\,(q+1)\,x^{m-\text{Mod}[m,n]}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p-\right.\right.$$
$$\left.\left.(-d)^{(m-\text{Mod}[m,n])/n-1}\,\left(c\,d^2-b\,d\,e+a\,e^2\right)^p\,(d\,(\text{Mod}[m,\,n]+1)+e\,(\text{Mod}[m,\,n]+n\,(q+1)+1)\,x^n)\right)\right)dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  (-d)^((m-Mod[m,n])/n-1)*(c*d^2-b*d*e+a*e^2)^p*x^(Mod[m,n]+1)*(d+e*x^n)^(q+1)/(n*e^(2*p+(m-Mod[m,n])/n)*(q+1)) +
  1/(n*e^(2*p+(m-Mod[m,n])/n)*(q+1))*Int[x^Mod[m,n]*(d+e*x^n)^(q+1)*
    ExpandToSum[Together[1/(d+e*x^n)*(n*e^(2*p+(m-Mod[m,n])/n)*(q+1)*x^(m-Mod[m,n])*(a+b*x^n+c*x^(2*n))^p-
      (-d)^((m-Mod[m,n])/n-1)*(c*d^2-b*d*e+a*e^2)^p*(d*(Mod[m,n]+1)+e*(Mod[m,n]+n*(q+1)+1)*x^n))],x],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && IGtQ[p,0] && ILtQ[q,-1] && IGtQ[m,0]
```

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  (-d)^((m-Mod[m,n])/n-1)*(c*d^2+a*e^2)^p*x^(Mod[m,n]+1)*(d+e*x^n)^(q+1)/(n*e^(2*p+(m-Mod[m,n])/n)*(q+1)) +
  1/(n*e^(2*p+(m-Mod[m,n])/n)*(q+1))*Int[x^Mod[m,n]*(d+e*x^n)^(q+1)*
    ExpandToSum[Together[1/(d+e*x^n)*(n*e^(2*p+(m-Mod[m,n])/n)*(q+1)*x^(m-Mod[m,n])*(a+c*x^(2*n))^p-
      (-d)^((m-Mod[m,n])/n-1)*(c*d^2+a*e^2)^p*(d*(Mod[m,n]+1)+e*(Mod[m,n]+n*(q+1)+1)*x^n))],x],x] /;
FreeQ[{a,c,d,e},x] && EqQ[n2,2*n] && IGtQ[n,0] && IGtQ[p,0] && ILtQ[q,-1] && IGtQ[m,0]
```

2: $\int x^m (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx$ when $b^2-4 a c \neq 0 \wedge (n \mid p) \in \mathbb{Z}^+ \wedge (m \mid q) \in \mathbb{Z} \wedge q<-1 \wedge m<0$

Derivation: Algebraic expansion and binomial recurrence 2b

Note: If $(n \mid p) \in \mathbb{Z}^+ \wedge (m \mid q) \in \mathbb{Z} \wedge q<0$**, then** $\frac{(-d)^{(m-\text{Mod}[m,n])/n}}{e^{2p+(m-\text{Mod}[m,n])/n}} \sum_{k=0}^{2p} (-d)^k e^{2p-k} P_{2p}[x^n, k]$ **is the coefficient of the** $x^{\text{Mod}[m,n]} (d+e x^n)^q$ **term of the partial fraction expansion of** $x^m P_{2p}[x^n] (d+e x^n)^q$**.**

Note: If $(n \mid p) \in \mathbb{Z}^+ \wedge (m \mid q) \in \mathbb{Z} \wedge q<-1 \wedge m<0$**, then** $n (-d)^{-(m-\text{Mod}[m,n])/n+1} e^{2p} (q+1) \left(a+b x^n+c x^{2n}\right)^p - e^{-(m-\text{Mod}[m,n])/n} \left(c d^2-b d e+a e^2\right)^p x^{-(m-\text{Mod}[m,n])} (d (\text{Mod}[m,n]+1)+e (\text{Mod}[m,n]+n (q+1)+1) x^n)$ **will be divisible by** $a+b x^n$**.**

Note: In the resulting integrand the degree of the polynomial in x^n **is at most** $q-1$**.**

Rule 1.2.3.4.6.1.1.1.2: If $b^2-4 a c \neq 0 \wedge (n \mid p) \in \mathbb{Z}^+ \wedge (m \mid q) \in \mathbb{Z} \wedge q<-1 \wedge m<0$**, then**

$$\int x^m (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx \longrightarrow$$

$$\frac{(-d)^{(m-\text{Mod}[m,n])/n}}{e^{2p+(m-\text{Mod}[m,n])/n}} \left(c d^2-b d e+a e^2\right)^p \int x^{\text{Mod}[m,n]} (d+e x^n)^q dx +$$
$$\frac{(-d)^{(m-\text{Mod}[m,n])/n}}{e^{2p}} \int x^m (d+e x^n)^q \left((-d)^{-(m-\text{Mod}[m,n])/n} e^{2p} \left(a+b x^n+c x^{2n}\right)^p - e^{-(m-\text{Mod}[m,n])/n} \left(c d^2-b d e+a e^2\right)^p x^{-m}\right) dx \longrightarrow$$

$$\frac{(-d)^{(m-\text{Mod}[m,n])/n-1} \left(c d^2-b d e+a e^2\right)^p x^{\text{Mod}[m,n]+1} (d+e x^n)^{q+1}}{n e^{2p+(m-\text{Mod}[m,n])/n} (q+1)} +$$

$$\frac{(-d)^{(m-\text{Mod}[m,n])/n-1}}{n\,e^{2p}\,(q+1)}\int x^m\,(d+e\,x^n)^{q+1}\cdot$$
$$\left(\frac{1}{d+e\,x^n}\left(n\,(-d)^{-(m-\text{Mod}[m,n])/n+1}\,e^{2p}\,(q+1)\,\left(a+b\,x^n+c\,x^{2n}\right)^p-\right.\right.$$
$$\left.\left.e^{-(m-\text{Mod}[m,n])/n}\left(c\,d^2-b\,d\,e+a\,e^2\right)^p x^{-(m-\text{Mod}[m,n])}\,(d\,(\text{Mod}[m,n]+1)+e\,(\text{Mod}[m,n]+n\,(q+1)+1)\,x^n)\right)\right)dx$$

Program code:

```
Int[x_^m_*(d_+e_.*x_^n_)^q_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  (-d)^((m-Mod[m,n])/n-1)*(c*d^2-b*d*e+a*e^2)^p*x^(Mod[m,n]+1)*(d+e*x^n)^(q+1)/(n*e^(2*p+(m-Mod[m,n])/n)*(q+1)) +
  (-d)^((m-Mod[m,n])/n-1)/(n*e^(2*p)*(q+1))*Int[x^m*(d+e*x^n)^(q+1)*
    ExpandToSum[Together[1/(d+e*x^n)*(n*(-d)^(-(m-Mod[m,n])/n+1)*e^(2*p)*(q+1)*(a+b*x^n+c*x^(2*n))^p -
  (e^(-(m-Mod[m,n])/n)*(c*d^2-b*d*e+a*e^2)^p*x^(-(m-Mod[m,n])))*(d*(Mod[m,n]+1)+e*(Mod[m,n]+n*(q+1)+1)*x^n))],x],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && IGtQ[p,0] && ILtQ[q,-1] && ILtQ[m,0]
```

```
Int[x_^m_*(d_+e_.*x_^n_)^q_*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  (-d)^((m-Mod[m,n])/n-1)*(c*d^2+a*e^2)^p*x^(Mod[m,n]+1)*(d+e*x^n)^(q+1)/(n*e^(2*p+(m-Mod[m,n])/n)*(q+1)) +
  (-d)^((m-Mod[m,n])/n-1)/(n*e^(2*p)*(q+1))*Int[x^m*(d+e*x^n)^(q+1)*
    ExpandToSum[Together[1/(d+e*x^n)*(n*(-d)^(-(m-Mod[m,n])/n+1)*e^(2*p)*(q+1)*(a+c*x^(2*n))^p -
  (e^(-(m-Mod[m,n])/n)*(c*d^2+a*e^2)^p*x^(-(m-Mod[m,n])))*(d*(Mod[m,n]+1)+e*(Mod[m,n]+n*(q+1)+1)*x^n))],x],x] /;
FreeQ[{a,c,d,e},x] && EqQ[n2,2*n] && IGtQ[n,0] && IGtQ[p,0] && IntegersQ[m,q] && ILtQ[q,-1] && ILtQ[m,0]
```

2: $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,(n\mid p)\in\mathbb{Z}^+\,\wedge\,2\,n\,p>n-1\,\wedge\,q\notin\mathbb{Z}\,\wedge\,m+2\,n\,p+n\,q+1\neq 0$

Reference: G&R 2.104

Note: This rule is a special case of the Ostrogradskiy-Hermite integration method.

Note: The degree of the polynomial in the resulting integrand is less than $2\,n$.

Rule 1.2.3.4.6.1.1.2: If $b^2-4\,a\,c\neq 0\,\wedge\,(n\mid p)\in\mathbb{Z}^+\,\wedge\,2\,n\,p>n-1\,\wedge\,q\notin\mathbb{Z}\,\wedge\,m+2\,n\,p+n\,q+1\neq 0$, then

$$\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2n}\right)^p dx\;\longrightarrow$$

$$\int (f\,x)^m\,(d+e\,x^n)^q\,\left(\left(a+b\,x^n+c\,x^{2n}\right)^p-x^{2np}\right)dx+\frac{c^p}{f^{2np}}\int (f\,x)^{m+2np}\,(d+e\,x^n)^q\,dx\;\longrightarrow$$

$$\frac{c^p\,(f\,x)^{m+2np-n+1}\,(d+e\,x^n)^{q+1}}{e\,f^{2np-n+1}\,(m+2\,n\,p+n\,q+1)}+$$

$$\frac{1}{e\,(m+2\,n\,p+n\,q+1)}\int (f\,x)^m\,(d+e\,x^n)^q\,\left(e\,(m+2\,n\,p+n\,q+1)\,\left(\left(a+b\,x^n+c\,x^{2\,n}\right)^p-c^p\,x^{2\,n\,p}\right)-d\,c^p\,(m+2\,n\,p-n+1)\,x^{2\,n\,p-n}\right)\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  c^p*(f*x)^(m+2*n*p-n+1)*(d+e*x^n)^(q+1)/(e*f^(2*n*p-n+1)*(m+2*n*p+n*q+1)) +
  1/(e*(m+2*n*p+n*q+1))*Int[(f*x)^m*(d+e*x^n)^q*
    ExpandToSum[e*(m+2*n*p+n*q+1)*((a+b*x^n+c*x^(2*n))^p-c^p*x^(2*n*p))-d*c^p*(m+2*n*p-n+1)*x^(2*n*p-n),x],x] /;
FreeQ[{a,b,c,d,e,f,m,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && IGtQ[p,0] && GtQ[2*n*p,n-1] &&
  Not[IntegerQ[q]] && NeQ[m+2*n*p+n*q+1,0]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  c^p*(f*x)^(m+2*n*p-n+1)*(d+e*x^n)^(q+1)/(e*f^(2*n*p-n+1)*(m+2*n*p+n*q+1)) +
  1/(e*(m+2*n*p+n*q+1))*Int[(f*x)^m*(d+e*x^n)^q*
    ExpandToSum[e*(m+2*n*p+n*q+1)*((a+c*x^(2*n))^p-c^p*x^(2*n*p))-d*c^p*(m+2*n*p-n+1)*x^(2*n*p-n),x],x] /;
FreeQ[{a,c,d,e,f,m,q},x] && EqQ[n2,2*n] && IGtQ[n,0] && IGtQ[p,0] && GtQ[2*n*p,n-1] &&
  Not[IntegerQ[q]] && NeQ[m+2*n*p+n*q+1,0]
```

3: $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \land\ (n\mid p)\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.3.4.6.1.1.3: If $b^2-4\,a\,c\neq 0\ \land\ n\in\mathbb{Z}^+\ \land\ p\in\mathbb{Z}^+$**, then**

$$\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\ \longrightarrow\ \int \text{ExpandIntegrand}\left[(f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p,\ x\right]\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m(d+e*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x],x] /;
FreeQ[{a,b,c,d,e,f,m,q},x] && EqQ[n2,2*n] && IGtQ[n,0] && IGtQ[p,0]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m(d+e*x^n)^q*(a+c*x^(2*n))^p,x],x] /;
FreeQ[{a,c,d,e,f,m,q},x] && EqQ[n2,2*n] && IGtQ[n,0] && IGtQ[p,0]
```

2: $\int x^m (d+e x^n)^q (a+b x^n+c x^{2n})^p \, dx$ when $b^2-4ac \neq 0 \land n \in \mathbb{Z}^+ \land m \in \mathbb{Z} \land \text{GCD}[m+1, n] \neq 1$

Derivation: Integration by substitution

- **Basis: If** $n \in \mathbb{Z} \land m \in \mathbb{Z}$, **let** $k = \text{GCD}[m+1, n]$, **then** $x^m F[x^n] = \frac{1}{k} \text{Subst}\left[x^{\frac{m+1}{k}-1} F\left[x^{n/k}\right], x, x^k\right] \partial_x x^k$

Rule 1.2.3.4.6.1.2: If $b^2-4ac \neq 0 \land n \in \mathbb{Z}^+ \land m \in \mathbb{Z}$, **let** $k = \text{GCD}[m+1, n]$, **if** $k \neq 1$, **then**

$$\int x^m (d+e x^n)^q (a+b x^n+c x^{2n})^p \, dx \longrightarrow \frac{1}{k} \text{Subst}\left[\int x^{\frac{m+1}{k}-1} \left(d+e x^{n/k}\right)^q \left(a+b x^{n/k}+c x^{2n/k}\right)^p dx, x, x^k\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{k=GCD[m+1,n]},
  1/k*Subst[Int[x^((m+1)/k-1)*(d+e*x^(n/k))^q*(a+b*x^(n/k)+c*x^(2*n/k))^p,x],x,x^k] /;
 k≠1] /;
FreeQ[{a,b,c,d,e,p,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && IntegerQ[m]
```

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{k=GCD[m+1,n]},
  1/k*Subst[Int[x^((m+1)/k-1)*(d+e*x^(n/k))^q*(a+c*x^(2*n/k))^p,x],x,x^k] /;
 k≠1] /;
FreeQ[{a,c,d,e,p,q},x] && EqQ[n2,2*n] && IGtQ[n,0] && IntegerQ[m]
```

3: $\int (f\,x)^m \left(d+e\,x^n\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\;\wedge\;n\in\mathbb{Z}^+\;\wedge\;m\in\mathbb{F}$

Derivation: Integration by substitution

- **Basis: If** $k\in\mathbb{Z}^+$**, then** $(f\,x)^m\,F[x]=\frac{k}{f}\,\text{Subst}\left[x^{k\,(m+1)-1}\,F\left[\frac{x^k}{f}\right],\;x,\;(f\,x)^{1/k}\right]\partial_x\,(f\,x)^{1/k}$

Rule 1.2.3.4.6.1.3: If $b^2-4\,a\,c\neq 0\;\wedge\;n\in\mathbb{Z}^+\;\wedge\;m\in\mathbb{F}$**, let** $k=\text{Denominator}[m]$**, then**

$$\int (f\,x)^m \left(d+e\,x^n\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx \;\longrightarrow\; \frac{k}{f}\,\text{Subst}\left[\int x^{k\,(m+1)-1}\left(d+\frac{e\,x^{k\,n}}{f^n}\right)^q\left(a+\frac{b\,x^{k\,n}}{f^n}+\frac{c\,x^{2\,k\,n}}{f^{2\,n}}\right)^p dx,\;x,\;(f\,x)^{1/k}\right]$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{k=Denominator[m]},
  k/f*Subst[Int[x^(k*(m+1)-1)*(d+e*x^(k*n)/f^n)^q*(a+b*x^(k*n)/f^n+c*x^(2*k*n)/f^(2*n))^p,x],x,(f*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e,f,p,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && FractionQ[m] && IntegerQ[p]
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{k=Denominator[m]},
  k/f*Subst[Int[x^(k*(m+1)-1)*(d+e*x^(k*n)/f)^q*(a+c*x^(2*k*n)/f)^p,x],x,(f*x)^(1/k)]] /;
FreeQ[{a,c,d,e,f,p,q},x] && EqQ[n2,2*n] && IGtQ[n,0] && FractionQ[m] && IntegerQ[p]
```

4. $\int (f\,x)^m \left(d+e\,x^n\right)\left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\;\wedge\;n\in\mathbb{Z}^+$

  1. $\int (f\,x)^m \left(d+e\,x^n\right)\left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\;\wedge\;n\in\mathbb{Z}^+\;\wedge\;p>0$

    1: $\int (f\,x)^m \left(d+e\,x^n\right)\left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\;\wedge\;n\in\mathbb{Z}^+\;\wedge\;p>0\;\wedge\;m<-1\;\wedge\;m+n\,(2\,p+1)+1\neq 0$

Derivation: Trinomial recurrence 1a

Rule 1.2.3.4.6.1.4.1.1: If $b^2-4\,a\,c\neq 0\;\wedge\;n\in\mathbb{Z}^+\;\wedge\;p>0\;\wedge\;m<-1\;\wedge\;m+n\,(2\,p+1)+1\neq 0$**, then**

$$\int (f\,x)^m \left(d+e\,x^n\right)\left(a+b\,x^n+c\,x^{2\,n}\right)^p dx \;\longrightarrow$$
$$\frac{(f\,x)^{m+1}\left(a+b\,x^n+c\,x^{2\,n}\right)^p\left(d\,(2\,n\,p+n+m+1)+e\,(m+1)\,x^n\right)}{f\,(m+1)\,(m+n\,(2\,p+1)+1)}+$$
$$\frac{n\,p}{f^n\,(m+1)\,(m+n\,(2\,p+1)+1)}\int (f\,x)^{m+n}\left(a+b\,x^n+c\,x^{2\,n}\right)^{p-1}\cdot$$

$$\left(2\,a\,e\,(m+1)-b\,d\,(m+n\,(2\,p+1)+1)+(b\,e\,(m+1)-2\,c\,d\,(m+n\,(2\,p+1)+1))\,x^n\right)\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)*(a_+b_.*x_^n_+c_.*x_^n2_)^p_.,x_Symbol] :=
  (f*x)^(m+1)*(a+b*x^n+c*x^(2*n))^p*(d*(m+n*(2*p+1)+1)+e*(m+1)*x^n)/(f*(m+1)*(m+n*(2*p+1)+1)) +
  n*p/(f^n*(m+1)*(m+n*(2*p+1)+1))*Int[(f*x)^(m+n)*(a+b*x^n+c*x^(2*n))^(p-1)*
      Simp[2*a*e*(m+1)-b*d*(m+n*(2*p+1)+1)+(b*e*(m+1)-2*c*d*(m+n*(2*p+1)+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GtQ[p,0] && LtQ[m,-1] && NeQ[m+n*(2*p+1)+1,0] && Integ
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)*(a_+c_.*x_^n2_)^p_.,x_Symbol] :=
  (f*x)^(m+1)*(a+c*x^(2*n))^p*(d*(m+n*(2*p+1)+1)+e*(m+1)*x^n)/(f*(m+1)*(m+n*(2*p+1)+1)) +
  2*n*p/(f^n*(m+1)*(m+n*(2*p+1)+1))*Int[(f*x)^(m+n)*(a+c*x^(2*n))^(p-1)*(a*e*(m+1)-c*d*(m+n*(2*p+1)+1)*x^n),x] /;
FreeQ[{a,c,d,e,f},x] && EqQ[n2,2*n] && IGtQ[n,0] && GtQ[p,0] && LtQ[m,-1] && NeQ[m+n*(2*p+1)+1,0] && IntegerQ[p]
```

2: $\int (f\,x)^m\,(d+e\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{Z}^+\,\land\,p>0\,\land\,m+2\,n\,p+1\neq 0\,\land\,m+n\,(2\,p+1)+1\neq 0$

Derivation: Trinomial recurrence 1b

Rule 1.2.3.4.6.1.4.1.2: If $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{Z}^+\,\land\,p>0\,\land\,m+2\,n\,p+1\neq 0\,\land\,m+n\,(2\,p+1)+1\neq 0$**, then**

$$\int (f\,x)^m\,(d+e\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow$$
$$\left((f\,x)^{m+1}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,\left(b\,e\,n\,p+c\,d\,(m+2\,n\,p+n+1)+c\,e\,(2\,n\,p+m+1)\,x^n\right)\right)\Big/\left(c\,f\,(m+2\,n\,p+1)\,(m+n\,(2\,p+1)+1)\right)+$$
$$\frac{n\,p}{c\,(m+2\,n\,p+1)\,(m+n\,(2\,p+1)+1)}\int (f\,x)^m\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p-1}\,\cdot$$
$$\left(2\,a\,c\,d\,(m+n\,(2\,p+1)+1)-a\,b\,e\,(m+1)+\left(2\,a\,c\,e\,(m+2\,n\,p+1)+b\,c\,d\,(m+n\,(2\,p+1)+1)-b^2\,e\,(m+n\,p+1)\right)\,x^n\right)\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)*(a_+b_.*x_^n_+c_.*x_^n2_)^p_.,x_Symbol] :=
  (f*x)^(m+1)*(a+b*x^n+c*x^(2*n))^p*(b*e*n*p+c*d*(m+n*(2*p+1)+1)+c*e*(2*n*p+m+1)*x^n)/
    (c*f*(2*n*p+m+1)*(m+n*(2*p+1)+1)) +
  n*p/(c*(2*n*p+m+1)*(m+n*(2*p+1)+1))*Int[(f*x)^m*(a+b*x^n+c*x^(2*n))^(p-1)*
    Simp[2*a*c*d*(m+n*(2*p+1)+1)-a*b*e*(m+1)+(2*a*c*e*(2*n*p+m+1)+b*c*d*(m+n*(2*p+1)+1)-b^2*e*(m+n*p+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GtQ[p,0] && NeQ[2*n*p+m+1,0] && NeQ[m+n*(2*p+1)+1,0]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)*(a_+c_.*x_^n2_)^p_.,x_Symbol] :=
  (f*x)^(m+1)*(a+c*x^(2*n))^p*(c*d*(m+n*(2*p+1)+1)+c*e*(2*n*p+m+1)*x^n)/(c*f*(2*n*p+m+1)*(m+n*(2*p+1)+1)) +
  2*a*n*p/((2*n*p+m+1)*(m+n*(2*p+1)+1))*Int[(f*x)^m*(a+c*x^(2*n))^(p-1)*Simp[d*(m+n*(2*p+1)+1)+e*(2*n*p+m+1)*x^n,x],x] /;
FreeQ[{a,c,d,e,f,m},x] && EqQ[n2,2*n] && IGtQ[n,0] && GtQ[p,0] && NeQ[2*n*p+m+1,0] && NeQ[m+n*(2*p+1)+1,0] && IntegerQ[p]
```

2. $\int (f\,x)^m\,(d+e\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1$

1: $\int (f\,x)^m\,(d+e\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1\,\wedge\,m>n-1$

Derivation: Trinomial recurrence 2a

Rule 1.2.3.4.6.1.4.2.1: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1\,\wedge\,m>n-1$**, then**

$$\int (f\,x)^m\,(d+e\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\longrightarrow$$

$$\frac{f^{n-1}\,(f\,x)^{m-n+1}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}\,(b\,d-2\,a\,e-(b\,e-2\,c\,d)\,x^n)}{n\,(p+1)\,\left(b^2-4\,a\,c\right)}+$$

$$\frac{f^n}{n\,(p+1)\,\left(b^2-4\,a\,c\right)}\int (f\,x)^{m-n}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}\,((n-m-1)\,(b\,d-2\,a\,e)+(2\,n\,p+2\,n+m+1)\,(b\,e-2\,c\,d)\,x^n)\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)*(a_+b_.*x_^n_+c_.*x_^n2_)^p_.,x_Symbol] :=
  f^(n-1)*(f*x)^(m-n+1)*(a+b*x^n+c*x^(2*n))^(p+1)*(b*d-2*a*e-(b*e-2*c*d)*x^n)/(n*(p+1)*(b^2-4*a*c)) +
  f^n/(n*(p+1)*(b^2-4*a*c))*Int[(f*x)^(m-n)*(a+b*x^n+c*x^(2*n))^(p+1)*
      Simp[(n-m-1)*(b*d-2*a*e)+(2*n*p+2*n+m+1)*(b*e-2*c*d)*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[p,-1] && GtQ[m,n-1] && IntegerQ[p]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)*(a_+c_.*x_^n2_)^p_.,x_Symbol] :=
  f^(n-1)*(f*x)^(m-n+1)*(a+c*x^(2*n))^(p+1)*(a*e-c*d*x^n)/(2*a*c*n*(p+1)) +
  f^n/(2*a*c*n*(p+1))*Int[(f*x)^(m-n)*(a+c*x^(2*n))^(p+1)*(a*e*(n-m-1)+c*d*(2*n*p+2*n+m+1)*x^n),x] /;
FreeQ[{a,c,d,e,f},x] && EqQ[n2,2*n] && IGtQ[n,0] && LtQ[p,-1] && GtQ[m,n-1] && IntegerQ[p]
```

2: $\int (f\,x)^m\,(d+e\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1$

Derivation: Trinomial recurrence 2b

Rule 1.2.3.4.6.1.4.2.2: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1$**, then**

$$\int (f\,x)^m\,(d+e\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\longrightarrow$$

$$-\left((f\,x)^{m+1}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}\,\left(d\,\left(b^2-2\,a\,c\right)-a\,b\,e+(b\,d-2\,a\,e)\,c\,x^n\right)\right)\Big/\left(a\,f\,n\,(p+1)\,\left(b^2-4\,a\,c\right)\right)+$$

$$\frac{1}{a\,n\,(p+1)\,\left(b^2-4\,a\,c\right)}\int(f\,x)^m\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}\cdot$$
$$\left(d\,\left(b^2\,(m+n\,(p+1)+1)-2\,a\,c\,(m+2\,n\,(p+1)+1)\right)-a\,b\,e\,(m+1)+c\,(m+n\,(2\,p+3)+1)\,(b\,d-2\,a\,e)\,x^n\right)\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  -(f*x)^(m+1)*(a+b*x^n+c*x^(2*n))^(p+1)*(d*(b^2-2*a*c)-a*b*e+(b*d-2*a*e)*c*x^n)/(a*f*n*(p+1)*(b^2-4*a*c)) +
  1/(a*n*(p+1)*(b^2-4*a*c))*Int[(f*x)^m*(a+b*x^n+c*x^(2*n))^(p+1)*
       Simp[d*(b^2*(m+n*(p+1)+1)-2*a*c*(m+2*n*(p+1)+1))-a*b*e*(m+1)+c*(m+n*(2*p+3)+1)*(b*d-2*a*e)*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[p,-1] && IntegerQ[p]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)*(a_+c_.*x_^n2_)^p_,x_Symbol] :=
  -(f*x)^(m+1)*(a+c*x^(2*n))^(p+1)*(d+e*x^n)/(2*a*f*n*(p+1)) +
  1/(2*a*n*(p+1))*Int[(f*x)^m*(a+c*x^(2*n))^(p+1)*Simp[d*(m+2*n*(p+1)+1)+e*(m+n*(2*p+3)+1)*x^n,x],x] /;
FreeQ[{a,c,d,e,f,m},x] && EqQ[n2,2*n] && IGtQ[n,0] && LtQ[p,-1] && IntegerQ[p]
```

3: $\int(f\,x)^m\,(d+e\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,m>n-1\,\wedge\,m+n\,(2\,p+1)+1\neq 0$

Derivation: Trinomial recurrence 3a

Rule 1.2.3.4.6.1.4.3: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,m>n-1\,\wedge\,m+n\,(2\,p+1)+1\neq 0$**, then**

$$\int(f\,x)^m\,(d+e\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow$$
$$\frac{e\,f^{n-1}\,(f\,x)^{m-n+1}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}}{c\,(m+n\,(2\,p+1)+1)}-$$
$$\frac{f^n}{c\,(m+n\,(2\,p+1)+1)}\int(f\,x)^{m-n}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,(a\,e\,(m-n+1)+(b\,e\,(m+n\,p+1)-c\,d\,(m+n\,(2\,p+1)+1))\,x^n)\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  e*f^(n-1)*(f*x)^(m-n+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(c*(m+n(2*p+1)+1)) -
  f^n/(c*(m+n(2*p+1)+1))*
    Int[(f*x)^(m-n)*(a+b*x^n+c*x^(2*n))^p*Simp[a*e*(m-n+1)+(b*e*(m+n*p+1)-c*d*(m+n(2*p+1)+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GtQ[m,n-1] && NeQ[m+n(2*p+1)+1,0] && IntegerQ[p]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)*(a_+c_.*x_^n2_)^p_,x_Symbol] :=
  e*f^(n-1)*(f*x)^(m-n+1)*(a+c*x^(2*n))^(p+1)/(c*(m+n(2*p+1)+1)) -
  f^n/(c*(m+n(2*p+1)+1))*Int[(f*x)^(m-n)*(a+c*x^(2*n))^p*(a*e*(m-n+1)-c*d*(m+n(2*p+1)+1)*x^n),x] /;
FreeQ[{a,c,d,e,f,p},x] && EqQ[n2,2*n] && IGtQ[n,0] && GtQ[m,n-1] && NeQ[m+n(2*p+1)+1,0] && IntegerQ[p]
```

4: $\int (f\,x)^m\,(d+e\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,m<-1$

Derivation: Trinomial recurrence 3b

Rule 1.2.3.4.6.1.4.4: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,m<-1$**, then**

$$\int (f\,x)^m\,(d+e\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow$$

$$\frac{d\,(f\,x)^{m+1}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}}{a\,f\,(m+1)}+$$

$$\frac{1}{a\,f^n\,(m+1)}\int (f\,x)^{m+n}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,\left(a\,e\,(m+1)-b\,d\,(m+n\,(p+1)+1)-c\,d\,(m+2\,n\,(p+1)+1)\,x^n\right)\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  d*(f*x)^(m+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(a*f*(m+1)) +
  1/(a*f^n*(m+1))*Int[(f*x)^(m+n)*(a+b*x^n+c*x^(2*n))^p*Simp[a*e*(m+1)-b*d*(m+n*(p+1)+1)-c*d*(m+2*n(p+1)+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[m,-1] && IntegerQ[p]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)*(a_+c_.*x_^n2_)^p_,x_Symbol] :=
  d*(f*x)^(m+1)*(a+c*x^(2*n))^(p+1)/(a*f*(m+1)) +
  1/(a*f^n*(m+1))*Int[(f*x)^(m+n)*(a+c*x^(2*n))^p*(a*e*(m+1)-c*d*(m+2*n(p+1)+1)*x^n),x] /;
FreeQ[{a,c,d,e,f,p},x] && EqQ[n2,2*n] && IGtQ[n,0] && LtQ[m,-1] && IntegerQ[p]
```

5. $\int \frac{(f\,x)^m\,(d+e\,x^n)}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq0\,\wedge\,n\in\mathbb{Z}^+$

1: $\int \frac{(f\,x)^m\,(d+e\,x^n)}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c<0\,\bigwedge\,\frac{n}{2}\in\mathbb{Z}^+\,\bigwedge\,0<m<n\,\bigwedge\,a\,c>0$

Derivation: Algebraic expansion

- Basis: Let $q=\sqrt{a\,c}$ and $r=\sqrt{2\,c\,q-b\,c}$, then $\frac{d+e\,z^2}{a+b\,z^2+c\,z^4}=\frac{c}{2\,q\,r}\,\frac{d\,r-(c\,d-e\,q)\,z}{q-r\,z+c\,z^2}+\frac{c}{2\,q\,r}\,\frac{d\,r+(c\,d-e\,q)\,z}{q+r\,z+c\,z^2}$
- Rule 1.2.3.4.6.1.4.5.1: If $b^2-4\,a\,c<0\,\bigwedge\,\frac{n}{2}\in\mathbb{Z}^+\,\bigwedge\,0<m<n\,\bigwedge\,a\,c>0$, let $q=\sqrt{a\,c}$, if $2\,c\,q-b\,c>0$, let $r=\sqrt{2\,c\,q-b\,c}$, then

$$\int \frac{(f\,x)^m\,(d+e\,x^n)}{a+b\,x^n+c\,x^{2\,n}}\,dx\,\longrightarrow\,\frac{c}{2\,q\,r}\int\frac{(f\,x)^m\,\left(d\,r-(c\,d-e\,q)\,x^{n/2}\right)}{q-r\,x^{n/2}+c\,x^n}\,dx+\frac{c}{2\,q\,r}\int\frac{(f\,x)^m\,\left(d\,r+(c\,d-e\,q)\,x^{n/2}\right)}{q+r\,x^{n/2}+c\,x^n}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^n_)/(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[a*c,2]},
  With[{r=Rt[2*c*q-b*c,2]},
  c/(2*q*r)*Int[(f*x)^m*Simp[d*r-(c*d-e*q)*x^(n/2),x]/(q-r*x^(n/2)+c*x^n),x] +
  c/(2*q*r)*Int[(f*x)^m*Simp[d*r+(c*d-e*q)*x^(n/2),x]/(q+r*x^(n/2)+c*x^n),x]] /;
 Not[LtQ[2*c*q-b*c,0]]] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[n2,2*n] && LtQ[b^2-4*a*c,0] && IntegersQ[m,n/2] && LtQ[0,m,n] && PosQ[a*c]
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^n_)/(a_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[a*c,2]},
  With[{r=Rt[2*c*q,2]},
  c/(2*q*r)*Int[(f*x)^m*Simp[d*r-(c*d-e*q)*x^(n/2),x]/(q-r*x^(n/2)+c*x^n),x] +
  c/(2*q*r)*Int[(f*x)^m*Simp[d*r+(c*d-e*q)*x^(n/2),x]/(q+r*x^(n/2)+c*x^n),x]] /;
 Not[LtQ[2*c*q,0]]] /;
FreeQ[{a,c,d,e,f},x] && EqQ[n2,2*n] && GtQ[a*c,0] && IntegersQ[m,n/2] && LtQ[0,m,n]
```

2: $\int \frac{(f\,x)^m\,(d+e\,x^n)}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c<0\,\bigwedge\,\frac{n}{2}-1\in\mathbb{Z}^+\,\bigwedge\,a\,c>0$

Derivation: Algebraic expansion

- Basis: Let $q=\sqrt{a\,c}$ and $r=\sqrt{2\,c\,q-b\,c}$, then $\frac{d+e\,z^2}{a+b\,z^2+c\,z^4}==\frac{c}{2\,q\,r}\,\frac{d\,r-(c\,d-e\,q)\,z}{q-r\,z+c\,z^2}+\frac{c}{2\,q\,r}\,\frac{d\,r+(c\,d-e\,q)\,z}{q+r\,z+c\,z^2}$
- Rule 1.2.3.4.6.1.4.5.2: If $b^2-4\,a\,c<0\,\bigwedge\,\frac{n}{2}-1\in\mathbb{Z}^+\,\bigwedge\,a\,c>0$, let $q=\sqrt{a\,c}$, if $2\,c\,q-b\,c>0$, let $r=\sqrt{2\,c\,q-b\,c}$, then

$$\int \frac{(f x)^m (d + e x^n)}{a + b x^n + c x^{2 n}} \, dx \longrightarrow \frac{c}{2 q r} \int \frac{(f x)^m \left(d r - (c d - e q) x^{n/2}\right)}{q - r x^{n/2} + c x^n} \, dx + \frac{c}{2 q r} \int \frac{(f x)^m \left(d r + (c d - e q) x^{n/2}\right)}{q + r x^{n/2} + c x^n} \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)/(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[a*c,2]},
  With[{r=Rt[2*c*q-b*c,2]},
  c/(2*q*r)*Int[(f*x)^m*(d*r-(c*d-e*q)*x^(n/2))/(q-r*x^(n/2)+c*x^n),x] +
  c/(2*q*r)*Int[(f*x)^m*(d*r+(c*d-e*q)*x^(n/2))/(q+r*x^(n/2)+c*x^n),x]] /;
 Not[LtQ[2*c*q-b*c,0]]] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[n2,2*n] && LtQ[b^2-4*a*c,0] && IGtQ[n/2,1] && PosQ[a*c]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)/(a_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[a*c,2]},
  With[{r=Rt[2*c*q,2]},
  c/(2*q*r)*Int[(f*x)^m*(d*r-(c*d-e*q)*x^(n/2))/(q-r*x^(n/2)+c*x^n),x] +
  c/(2*q*r)*Int[(f*x)^m*(d*r+(c*d-e*q)*x^(n/2))/(q+r*x^(n/2)+c*x^n),x]] /;
 Not[LtQ[2*c*q,0]]] /;
FreeQ[{a,c,d,e,f,m},x] && EqQ[n2,2*n] && IGtQ[n/2,1] && GtQ[a*c,0]
```

3: $\int \frac{(f x)^m (d + e x^n)}{a + b x^n + c x^{2 n}} \, dx$ when $b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

- **Basis: Let** $q \to \sqrt{b^2 - 4 a c}$ **, then** $\frac{d+e z}{a+b z+c z^2} = \left(\frac{e}{2} + \frac{2 c d-b e}{2 q}\right) \frac{1}{\frac{b}{2}-\frac{q}{2}+c z} + \left(\frac{e}{2} - \frac{2 c d-b e}{2 q}\right) \frac{1}{\frac{b}{2}+\frac{q}{2}+c z}$

- **Rule 1.2.3.4.6.1.4.5.3: If** $b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}^+$**, let** $q \to \sqrt{b^2 - 4 a c}$**, then**

$$\int \frac{(f x)^m (d + e x^n)}{a + b x^n + c x^{2 n}} \, dx \longrightarrow \left(\frac{e}{2} + \frac{2 c d - b e}{2 q}\right) \int \frac{(f x)^m}{\frac{b}{2} - \frac{q}{2} + c x^n} \, dx + \left(\frac{e}{2} - \frac{2 c d - b e}{2 q}\right) \int \frac{(f x)^m}{\frac{b}{2} + \frac{q}{2} + c x^n} \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)/(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (e/2+(2*c*d-b*e)/(2*q))*Int[(f*x)^m/(b/2-q/2+c*x^n),x] + (e/2-(2*c*d-b*e)/(2*q))*Int[(f*x)^m/(b/2+q/2+c*x^n),x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)/(a_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[-a*c,2]},
  -(e/2+c*d/(2*q))*Int[(f*x)^m/(q-c*x^n),x] + (e/2-c*d/(2*q))*Int[(f*x)^m/(q+c*x^n),x]] /;
FreeQ[{a,c,d,e,f,m},x] && EqQ[n2,2*n] && IGtQ[n,0]
```

5. $\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+$

  1. $\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,q\in\mathbb{Z}$

    1: $\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,q\in\mathbb{Z}\,\wedge\,m\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule 1.2.3.4.6.1.5.1.1: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,q\in\mathbb{Z}\,\wedge\,m\in\mathbb{Z}$**, then**

$$\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx\;\longrightarrow\;\int \text{ExpandIntegrand}\left[\frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}},\,x\right]dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_./(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m*(d+e*x^n)^q/(a+b*x^n+c*x^(2*n)),x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && IntegerQ[q] && IntegerQ[m]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_./(a_+c_.*x_^n2_.),x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m*(d+e*x^n)^q/(a+c*x^(2*n)),x],x] /;
FreeQ[{a,c,d,e,f,m},x] && EqQ[n2,2*n] && IGtQ[n,0] && IntegerQ[q] && IntegerQ[m]
```

2: $\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,q\in\mathbb{Z}\,\wedge\,m\notin\mathbb{Z}$

Derivation: Algebraic expansion

Rule 1.2.3.4.6.1.5.1.2: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,q\in\mathbb{Z}\,\wedge\,m\notin\mathbb{Z}$**, then**

$$\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx\;\longrightarrow\;\int (f\,x)^m\,\text{ExpandIntegrand}\left[\frac{(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}},\,x\right]dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_./(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m,(d+e*x^n)^q/(a+b*x^n+c*x^(2*n)),x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && IntegerQ[q] && Not[IntegerQ[m]]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_./(a_+c_.*x_^n2_.),x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m,(d+e*x^n)^q/(a+c*x^(2*n)),x],x] /;
FreeQ[{a,c,d,e,f,m},x] && EqQ[n2,2*n] && IGtQ[n,0] && IntegerQ[q] && Not[IntegerQ[m]]
```

2. $\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,q\notin\mathbb{Z}$

1. $\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,q\notin\mathbb{Z}\,\wedge\,q>0$

1. $\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,q\notin\mathbb{Z}\,\wedge\,q>0\,\wedge\,m>n-1$

1: $\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,q\notin\mathbb{Z}\,\wedge\,q>0\,\wedge\,m>2\,n-1$

Reference: Algebraic expansion

Basis: $\frac{d+e\,z}{a+b\,z+c\,z^2} == \frac{c\,d-b\,e+c\,e\,z}{c^2\,z^2} - \frac{a\,(c\,d-b\,e)+(b\,c\,d-b^2\,e+a\,c\,e)\,z}{c^2\,z^2\,(a+b\,z+c\,z^2)}$

Rule 1.2.3.4.6.1.5.2.1.1.1: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,q\notin\mathbb{Z}\,\wedge\,q>0\,\wedge\,m>2\,n-1$**, then**

$$\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx \longrightarrow$$

$$\frac{f^{2\,n}}{c^2}\int (f\,x)^{m-2\,n}\,(c\,d-b\,e+c\,e\,x^n)\,(d+e\,x^n)^{q-1}\,dx - \frac{f^{2\,n}}{c^2}\int \frac{(f\,x)^{m-2\,n}\,(d+e\,x^n)^{q-1}\,\left(a\,(c\,d-b\,e)+\left(b\,c\,d-b^2\,e+a\,c\,e\right)\,x^n\right)}{a+b\,x^n+c\,x^{2\,n}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^n_)^q_/(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  f^(2*n)/c^2*Int[(f*x)^(m-2*n)*(c*d-b*e+c*e*x^n)*(d+e*x^n)^(q-1),x] -
  f^(2*n)/c^2*Int[(f*x)^(m-2*n)*(d+e*x^n)^(q-1)*Simp[a*(c*d-b*e)+(b*c*d-b^2*e+a*c*e)*x^n,x]/(a+b*x^n+c*x^(2*n)),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && Not[IntegerQ[q]] && GtQ[q,0] && GtQ[m,2*n-1]
```

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^n_)^q_/(a_+c_.*x_^n2_.),x_Symbol] :=
  f^(2*n)/c*Int[(f*x)^(m-2*n)*(d+e*x^n)^q,x] -
  a*f^(2*n)/c*Int[(f*x)^(m-2*n)*(d+e*x^n)^q/(a+c*x^(2*n)),x] /;
FreeQ[{a,c,d,e,f,q},x] && EqQ[n2,2*n] && IGtQ[n,0] && Not[IntegerQ[q]] && GtQ[m,2*n-1]
```

2: $\int \frac{(f x)^m (d+e x^n)^q}{a+b x^n+c x^{2 n}} dx$ when $b^2-4 a c \neq 0 \wedge n \in \mathbb{Z}^+ \wedge q \notin \mathbb{Z} \wedge q>0 \wedge n-1<m \leq 2 n-1$

Reference: Algebraic expansion

Basis: $\frac{d+e z}{a+b z+c z^2} = \frac{e}{c z} - \frac{a e-(c d-b e) z}{c z (a+b z+c z^2)}$

Rule 1.2.3.4.6.1.5.2.1.1.2: If $b^2-4 a c \neq 0 \wedge n \in \mathbb{Z}^+ \wedge q \notin \mathbb{Z} \wedge q>0 \wedge n-1<m \leq 2 n-1$**, then**

$$\int \frac{(f x)^m (d+e x^n)^q}{a+b x^n+c x^{2 n}} dx \longrightarrow \frac{e f^n}{c} \int (f x)^{m-n} (d+e x^n)^{q-1} dx - \frac{f^n}{c} \int \frac{(f x)^{m-n} (d+e x^n)^{q-1} (a e-(c d-b e) x^n)}{a+b x^n+c x^{2 n}} dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^n_)^q_/(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  e*f^n/c*Int[(f*x)^(m-n)*(d+e*x^n)^(q-1),x] -
  f^n/c*Int[(f*x)^(m-n)*(d+e*x^n)^(q-1)*Simp[a*e-(c*d-b*e)*x^n,x]/(a+b*x^n+c*x^(2*n)),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && Not[IntegerQ[q]] && GtQ[q,0] && GtQ[m,n-1] && LeQ[m,2n-
```

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^n_)^q_/(a_+c_.*x_^n2_.),x_Symbol] :=
  e*f^n/c*Int[(f*x)^(m-n)*(d+e*x^n)^(q-1),x] -
  f^n/c*Int[(f*x)^(m-n)*(d+e*x^n)^(q-1)*Simp[a*e-c*d*x^n,x]/(a+c*x^(2*n)),x] /;
FreeQ[{a,c,d,e,f},x] && EqQ[n2,2*n] && IGtQ[n,0] && Not[IntegerQ[q]] && GtQ[q,0] && GtQ[m,n-1] && LeQ[m,2n-1]
```

2: $\int \frac{(f x)^m (d+e x^n)^q}{a+b x^n+c x^{2 n}} dx$ when $b^2-4 a c \neq 0 \wedge n \in \mathbb{Z}^+ \wedge q \notin \mathbb{Z} \wedge q>0 \wedge m<0$

Reference: Algebraic expansion

Basis: $\frac{d+e z}{a+b z+c z^2} = \frac{d}{a} - \frac{z (b d-a e+c d z)}{a (a+b z+c z^2)}$

Rule 1.2.3.4.6.1.5.2.1.2: If $b^2-4 a c \neq 0 \wedge n \in \mathbb{Z}^+ \wedge q \notin \mathbb{Z} \wedge q>0 \wedge m<0$**, then**

$$\int \frac{(f x)^m (d+e x^n)^q}{a+b x^n+c x^{2 n}} dx \longrightarrow \frac{d}{a} \int (f x)^m (d+e x^n)^{q-1} dx - \frac{1}{a f^n} \int \frac{(f x)^{m+n} (d+e x^n)^{q-1} (b d-a e+c d x^n)}{a+b x^n+c x^{2 n}} dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_.+e_.*x_^n_)^q_/(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  d/a*Int[(f*x)^m*(d+e*x^n)^(q-1),x] -
  1/(a*f^n)*Int[(f*x)^(m+n)*(d+e*x^n)^(q-1)*Simp[b*d-a*e+c*d*x^n,x]/(a+b*x^n+c*x^(2*n)),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && Not[IntegerQ[q]] && GtQ[q,0] && LtQ[m,0]
```

```
Int[(f_.*x_)^m_*(d_.+e_.*x_^n_)^q_/(a_+c_.*x_^n2_.),x_Symbol] :=
  d/a*Int[(f*x)^m*(d+e*x^n)^(q-1),x] +
  1/(a*f^n)*Int[(f*x)^(m+n)*(d+e*x^n)^(q-1)*Simp[a*e-c*d*x^n,x]/(a+c*x^(2*n)),x] /;
FreeQ[{a,c,d,e,f},x] && EqQ[n2,2*n] && IGtQ[n,0] && Not[IntegerQ[q]] && GtQ[q,0] && LtQ[m,0]
```

2. $\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,q\notin\mathbb{Z}\,\wedge\,q<-1$

   1. $\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,q\notin\mathbb{Z}\,\wedge\,q<-1\,\wedge\,m>n-1$

      1: $\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,q\notin\mathbb{Z}\,\wedge\,q<-1\,\wedge\,m>2\,n-1$

Reference: Algebraic expansion

- **Basis:** $\frac{1}{a+b\,z+c\,z^2}=\frac{d^2}{(c\,d^2-b\,d\,e+a\,e^2)\,z^2}-\frac{(d+e\,z)\,(a\,d+(b\,d-a\,e)\,z)}{(c\,d^2-b\,d\,e+a\,e^2)\,z^2\,(a+b\,z+c\,z^2)}$

Rule 1.2.3.4.6.1.5.2.2.1.1: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,q\notin\mathbb{Z}\,\wedge\,q<-1\,\wedge\,m>2\,n-1$**, then**

$$\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx\,\longrightarrow\,\frac{d^2\,f^{2\,n}}{c\,d^2-b\,d\,e+a\,e^2}\int (f\,x)^{m-2\,n}\,(d+e\,x^n)^q\,dx-\frac{f^{2\,n}}{c\,d^2-b\,d\,e+a\,e^2}\int \frac{(f\,x)^{m-2\,n}\,(d+e\,x^n)^{q+1}\,(a\,d+(b\,d-a\,e)\,x^n)}{a+b\,x^n+c\,x^{2\,n}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^n_)^q_/(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  d^2*f^(2*n)/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-2*n)*(d+e*x^n)^q,x] -
  f^(2*n)/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-2*n)*(d+e*x^n)^(q+1)*Simp[a*d+(b*d-a*e)*x^n,x]/(a+b*x^n+c*x^(2*n)),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && Not[IntegerQ[q]] && LtQ[q,-1] && GtQ[m,2*n-1]
```

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^n_)^q_/(a_+c_.*x_^n2_.),x_Symbol] :=
  d^2*f^(2*n)/(c*d^2+a*e^2)*Int[(f*x)^(m-2*n)*(d+e*x^n)^q,x] -
  a*f^(2*n)/(c*d^2+a*e^2)*Int[(f*x)^(m-2*n)*(d+e*x^n)^(q+1)*(d-e*x^n)/(a+c*x^(2*n)),x] /;
FreeQ[{a,c,d,e,f},x] && EqQ[n2,2*n] && IGtQ[n,0] && Not[IntegerQ[q]] && LtQ[q,-1] && GtQ[m,2*n-1]
```

2: $\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,q\notin\mathbb{Z}\,\wedge\,q<-1\,\wedge\,n-1<m\leq 2\,n-1$

Reference: Algebraic expansion

– **Basis:** $\frac{1}{a+b\,z+c\,z^2}==-\frac{d\,e}{(c\,d^2-b\,d\,e+a\,e^2)\,z}+\frac{(d+e\,z)\,(a\,e+c\,d\,z)}{(c\,d^2-b\,d\,e+a\,e^2)\,z\,(a+b\,z+c\,z^2)}$

Rule 1.2.3.4.6.1.5.2.2.1.2: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,q\notin\mathbb{Z}\,\wedge\,q<-1\,\wedge\,n-1<m\leq 2\,n-1$**, then**

$$\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx\,\longrightarrow\,-\frac{d\,e\,f^n}{c\,d^2-b\,d\,e+a\,e^2}\int (f\,x)^{m-n}\,(d+e\,x^n)^q\,dx+\frac{f^n}{c\,d^2-b\,d\,e+a\,e^2}\int \frac{(f\,x)^{m-n}\,(d+e\,x^n)^{q+1}\,(a\,e+c\,d\,x^n)}{a+b\,x^n+c\,x^{2\,n}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^n_)^q_/(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  -d*e*f^n/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-n)*(d+e*x^n)^q,x] +
  f^n/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-n)*(d+e*x^n)^(q+1)*Simp[a*e+c*d*x^n,x]/(a+b*x^n+c*x^(2*n)),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && Not[IntegerQ[q]] && LtQ[q,-1] && GtQ[m,n-1] && LeQ[m,2
```

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^n_)^q_/(a_+c_.*x_^n2_.),x_Symbol] :=
  -d*e*f^n/(c*d^2+a*e^2)*Int[(f*x)^(m-n)*(d+e*x^n)^q,x] +
  f^n/(c*d^2+a*e^2)*Int[(f*x)^(m-n)*(d+e*x^n)^(q+1)*Simp[a*e+c*d*x^n,x]/(a+c*x^(2*n)),x] /;
FreeQ[{a,c,d,e,f},x] && EqQ[n2,2*n] && IGtQ[n,0] && Not[IntegerQ[q]] && LtQ[q,-1] && GtQ[m,n-1] && LeQ[m,2*n-1]
```

2: $\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,q\notin\mathbb{Z}\,\wedge\,q<-1$

Derivation: Algebraic expansion

– **Basis:** $\frac{1}{a+b\,z+c\,z^2}==\frac{e^2}{c\,d^2-b\,d\,e+a\,e^2}+\frac{(d+e\,z)\,(c\,d-b\,e-c\,e\,z)}{(c\,d^2-b\,d\,e+a\,e^2)\,(a+b\,z+c\,z^2)}$

Rule 1.2.3.4.6.1.5.2.2.2: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,q\notin\mathbb{Z}\,\wedge\,q<-1$**, then**

$$\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx\,\longrightarrow\,\frac{e^2}{c\,d^2-b\,d\,e+a\,e^2}\int (f\,x)^m\,(d+e\,x^n)^q\,dx+\frac{1}{c\,d^2-b\,d\,e+a\,e^2}\int \frac{(f\,x)^m\,(d+e\,x^n)^{q+1}\,(c\,d-b\,e-c\,e\,x^n)}{a+b\,x^n+c\,x^{2\,n}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_/(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  e^2/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^m*(d+e*x^n)^q,x] +
  1/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^m*(d+e*x^n)^(q+1)*Simp[c*d-b*e-c*e*x^n,x]/(a+b*x^n+c*x^(2*n)),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && Not[IntegerQ[q]] && LtQ[q,-1]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_/(a_+c_.*x_^n2_),x_Symbol] :=
  e^2/(c*d^2+a*e^2)*Int[(f*x)^m*(d+e*x^n)^q,x] +
  c/(c*d^2+a*e^2)*Int[(f*x)^m*(d+e*x^n)^(q+1)*(d-e*x^n)/(a+c*x^(2*n)),x] /;
FreeQ[{a,c,d,e,f,m},x] && EqQ[n2,2*n] && IGtQ[n,0] && Not[IntegerQ[q]] && LtQ[q,-1]
```

3: $\int \frac{(f x)^m (d+e x^n)^q}{a+b x^n+c x^{2n}} dx$ when $b^2-4ac \neq 0 \land n \in \mathbb{Z}^+ \land q \notin \mathbb{Z} \land m \in \mathbb{Z}$

Derivation: Algebraic expansion

- **Basis: If** $q=\sqrt{b^2-4ac}$, **then** $\frac{1}{a+b z+c z^2} == \frac{2c}{q(b-q+2cz)} - \frac{2c}{q(b+q+2cz)}$

Rule 1.2.3.4.6.1.5.2.3: If $b^2-4ac \neq 0 \land n \in \mathbb{Z}^+ \land q \notin \mathbb{Z} \land m \in \mathbb{Z}$, **then**

$$\int \frac{(f x)^m (d+e x^n)^q}{a+b x^n+c x^{2n}} dx \longrightarrow \int (d+e x^n)^q \text{ExpandIntegrand}\left[\frac{(f x)^m}{a+b x^n+c x^{2n}}, x\right] dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_/(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^n)^q,(f*x)^m/(a+b*x^n+c*x^(2*n)),x],x] /;
FreeQ[{a,b,c,d,e,f,q,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && Not[IntegerQ[q]] && IntegerQ[m]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_/(a_+c_.*x_^n2_.),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^n)^q,(f*x)^m/(a+c*x^(2*n)),x],x] /;
FreeQ[{a,c,d,e,f,q,n},x] && EqQ[n2,2*n] && IGtQ[n,0] && Not[IntegerQ[q]] && IntegerQ[m]
```

4: $\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,q\notin\mathbb{Z}\,\wedge\,m\notin\mathbb{Z}$

Derivation: Algebraic expansion

- **Basis: If** $q=\sqrt{b^2-4\,a\,c}$, **then** $\frac{1}{a+b\,z+c\,z^2}==\frac{2\,c}{q\,(b-q+2\,c\,z)}-\frac{2\,c}{q\,(b+q+2\,c\,z)}$

Rule 1.2.3.4.6.1.5.2.4: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,q\notin\mathbb{Z}\,\wedge\,m\notin\mathbb{Z}$, **then**

$$\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx\,\longrightarrow\,\int (f\,x)^m\,(d+e\,x^n)^q\,\text{ExpandIntegrand}\left[\frac{1}{a+b\,x^n+c\,x^{2\,n}},\,x\right]dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_/(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m*(d+e*x^n)^q,1/(a+b*x^n+c*x^(2*n)),x],x] /;
FreeQ[{a,b,c,d,e,f,m,q,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && Not[IntegerQ[q]] && Not[IntegerQ[m]]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_/(a_+c_.*x_^n2_.),x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m*(d+e*x^n)^q,1/(a+c*x^(2*n)),x],x] /;
FreeQ[{a,c,d,e,f,m,q,n},x] && EqQ[n2,2*n] && IGtQ[n,0] && Not[IntegerQ[q]] && Not[IntegerQ[m]]
```

6. $\int \frac{(f\,x)^m\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{d+e\,x^n}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{Z}^+$

1. $\int \frac{(f\,x)^m\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{d+e\,x^n}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{Z}^+\land\,p>0\,\land\,m<0$

1: $\int \frac{(f\,x)^m\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{d+e\,x^n}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{Z}^+\land\,p>0\,\land\,m<-n$

Reference: Algebraic expansion

- **Basis:** $\frac{a+b\,z+c\,z^2}{d+e\,z}==\frac{a\,d+(b\,d-a\,e)\,z}{d^2}+\frac{\left(c\,d^2-b\,d\,e+a\,e^2\right)\,z^2}{d^2\,(d+e\,z)}$

Rule 1.2.3.4.6.1.6.1.1: If $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{Z}^+\land\,p>0\,\land\,m<-n$**, then**

$$\int \frac{(f\,x)^m\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{d+e\,x^n}\,dx\,\longrightarrow$$

$$\frac{1}{d^2}\int (f\,x)^m\,\left(a\,d+(b\,d-a\,e)\,x^n\right)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p-1}\,dx+\frac{c\,d^2-b\,d\,e+a\,e^2}{d^2\,f^{2\,n}}\int \frac{(f\,x)^{m+2\,n}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p-1}}{d+e\,x^n}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*x_^n_+c_.*x_^n2_.)^p_./(d_.+e_.*x_^n_),x_Symbol] :=
  1/d^2*Int[(f*x)^m*(a*d+(b*d-a*e)*x^n)*(a+b*x^n+c*x^(2*n))^(p-1),x] +
  (c*d^2-b*d*e+a*e^2)/(d^2*f^(2*n))*Int[(f*x)^(m+2*n)*(a+b*x^n+c*x^(2*n))^(p-1)/(d+e*x^n),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GtQ[p,0] && LtQ[m,-n]
```

```
Int[(f_.*x_)^m_*(a_+c_.*x_^n2_.)^p_./(d_.+e_.*x_^n_),x_Symbol] :=
  a/d^2*Int[(f*x)^m*(d-e*x^n)*(a+c*x^(2*n))^(p-1),x] +
  (c*d^2+a*e^2)/(d^2*f^(2*n))*Int[(f*x)^(m+2*n)*(a+c*x^(2*n))^(p-1)/(d+e*x^n),x] /;
FreeQ[{a,c,d,e,f},x] && EqQ[n2,2*n] && IGtQ[n,0] && GtQ[p,0] && LtQ[m,-n]
```

2: $\int \frac{(f\,x)^m\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{d+e\,x^n}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{Z}^+\land\,p>0\,\land\,m<0$

Reference: Algebraic expansion

- **Basis:** $\frac{a+b\,z+c\,z^2}{d+e\,z}==\frac{a\,e+c\,d\,z}{d\,e}-\frac{\left(c\,d^2-b\,d\,e+a\,e^2\right)\,z}{d\,e\,(d+e\,z)}$

Rule 1.2.3.4.6.1.6.1.2: If $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{Z}^+\land\,p>0\,\land\,m<0$**, then**

$$\int \frac{(f\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p}{d+e\,x^n}\,dx \longrightarrow$$

$$\frac{1}{d\,e}\int (f\,x)^m\,(a\,e+c\,d\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p-1}dx-\frac{c\,d^2-b\,d\,e+a\,e^2}{d\,e\,f^n}\int \frac{(f\,x)^{m+n}\left(a+b\,x^n+c\,x^{2\,n}\right)^{p-1}}{d+e\,x^n}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*x_^n_+c_.*x_^n2_.)^p_./(d_.+e_.*x_^n_),x_Symbol] :=
  1/(d*e)*Int[(f*x)^m*(a*e+c*d*x^n)*(a+b*x^n+c*x^(2*n))^(p-1),x] -
  (c*d^2-b*d*e+a*e^2)/(d*e*f^n)*Int[(f*x)^(m+n)*(a+b*x^n+c*x^(2*n))^(p-1)/(d+e*x^n),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GtQ[p,0] && LtQ[m,0]
```

```
Int[(f_.*x_)^m_*(a_+c_.*x_^n2_.)^p_./(d_.+e_.*x_^n_),x_Symbol] :=
  1/(d*e)*Int[(f*x)^m*(a*e+c*d*x^n)*(a+c*x^(2*n))^(p-1),x] -
  (c*d^2+a*e^2)/(d*e*f^n)*Int[(f*x)^(m+n)*(a+c*x^(2*n))^(p-1)/(d+e*x^n),x] /;
FreeQ[{a,c,d,e,f},x] && EqQ[n2,2*n] && IGtQ[n,0] && GtQ[p,0] && LtQ[m,0]
```

2. $\int \frac{(f\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p}{d+e\,x^n}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1\,\wedge\,m>0$

1: $\int \frac{(f\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p}{d+e\,x^n}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1\,\wedge\,m>n$

Reference: Algebraic expansion

- **Basis:** $\frac{z^2}{d+e\,z}=-\frac{a\,d+(b\,d-a\,e)\,z}{c\,d^2-b\,d\,e+a\,e^2}+\frac{d^2\,(a+b\,z+c\,z^2)}{(c\,d^2-b\,d\,e+a\,e^2)\,(d+e\,z)}$

Rule 1.2.3.4.6.1.6.2.1: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1\,\wedge\,m>n$**, then**

$$\int \frac{(f\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p}{d+e\,x^n}\,dx \longrightarrow$$

$$-\frac{f^{2\,n}}{c\,d^2-b\,d\,e+a\,e^2}\int (f\,x)^{m-2\,n}\,(a\,d+(b\,d-a\,e)\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p dx+\frac{d^2\,f^{2\,n}}{c\,d^2-b\,d\,e+a\,e^2}\int \frac{(f\,x)^{m-2\,n}\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}}{d+e\,x^n}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(a_.+b_.*x_^n_+c_.*x_^n2_.)^p_/(d_.+e_.*x_^n_),x_Symbol] :=
  -f^(2*n)/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-2*n)*(a*d+(b*d-a*e)*x^n)*(a+b*x^n+c*x^(2*n))^p,x] +
  d^2*f^(2*n)/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-2*n)*(a+b*x^n+c*x^(2*n))^(p+1)/(d+e*x^n),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[p,-1] && GtQ[m,n]
```

```
Int[(f_.*x_)^m_.*(a_+c_.*x_^n2_.)^p_/(d_.+e_.*x_^n_),x_Symbol] :=
  -a*f^(2*n)/(c*d^2+a*e^2)*Int[(f*x)^(m-2*n)*(d-e*x^n)*(a+c*x^(2*n))^p,x] +
  d^2*f^(2*n)/(c*d^2+a*e^2)*Int[(f*x)^(m-2*n)*(a+c*x^(2*n))^(p+1)/(d+e*x^n),x] /;
FreeQ[{a,c,d,e,f},x] && EqQ[n2,2*n] && IGtQ[n,0] && LtQ[p,-1] && GtQ[m,n]
```

2: $\int \frac{(f\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p}{d+e\,x^n}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,p<-1\,\wedge\,m>0$

Reference: Algebraic expansion

- **Basis:** $\frac{z}{d+e\,z}==\frac{a\,e+c\,d\,z}{c\,d^2-b\,d\,e+a\,e^2}-\frac{d\,e\,\left(a+b\,z+c\,z^2\right)}{\left(c\,d^2-b\,d\,e+a\,e^2\right)\,\left(d+e\,z\right)}$

Rule 1.2.3.4.6.1.6.2.2: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,p<-1\,\wedge\,m>0$**, then**

$$\int \frac{(f\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p}{d+e\,x^n}\,dx\,\longrightarrow$$

$$\frac{f^n}{c\,d^2-b\,d\,e+a\,e^2}\int (f\,x)^{m-n}\,\left(a\,e+c\,d\,x^n\right)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx-\frac{d\,e\,f^n}{c\,d^2-b\,d\,e+a\,e^2}\int \frac{(f\,x)^{m-n}\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}}{d+e\,x^n}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(a_.+b_.*x_^n_+c_.*x_^n2_.)^p_/(d_.+e_.*x_^n_),x_Symbol] :=
  f^n/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-n)*(a*e+c*d*x^n)*(a+b*x^n+c*x^(2*n))^p,x] -
  d*e*f^n/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-n)*(a+b*x^n+c*x^(2*n))^(p+1)/(d+e*x^n),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[p,-1] && GtQ[m,0]
```

```
Int[(f_.*x_)^m_.*(a_+c_.*x_^n2_.)^p_/(d_.+e_.*x_^n_),x_Symbol] :=
  f^n/(c*d^2+a*e^2)*Int[(f*x)^(m-n)*(a*e+c*d*x^n)*(a+c*x^(2*n))^p,x] -
  d*e*f^n/(c*d^2+a*e^2)*Int[(f*x)^(m-n)*(a+c*x^(2*n))^(p+1)/(d+e*x^n),x] /;
FreeQ[{a,c,d,e,f},x] && EqQ[n2,2*n] && IGtQ[n,0] && LtQ[p,-1] && GtQ[m,0]
```

7: $\int (f x)^m (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx$ when $b^2-4ac\neq 0 \wedge n\in\mathbb{Z}^+ \wedge (q\in\mathbb{Z}^+ \vee (m \mid q)\in\mathbb{Z})$

Derivation: Algebraic expansion

Rule 1.2.3.4.6.1.7: If $b^2-4ac\neq 0 \wedge n\in\mathbb{Z}^+ \wedge (q\in\mathbb{Z}^+ \vee (m \mid q)\in\mathbb{Z})$**, then**

$$\int (f x)^m (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx \longrightarrow \int \left(a+b x^n+c x^{2n}\right)^p \text{ExpandIntegrand}[(f x)^m (d+e x^n)^q, x]\, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x^n+c*x^(2*n))^p,(f*x)^m(d+e*x^n)^q,x],x] /;
FreeQ[{a,b,c,d,e,f,m,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && (IGtQ[q,0] || IntegersQ[m,q])
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+c*x^(2*n))^p,(f*x)^m(d+e*x^n)^q,x],x] /;
FreeQ[{a,c,d,e,f,m,q},x] && EqQ[n2,2*n] && IGtQ[n,0] && IGtQ[q,0]
```

2. $\int (f x)^m (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx$ when $b^2-4ac\neq 0 \wedge n\in\mathbb{Z}^-$

1. $\int (f x)^m (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx$ when $b^2-4ac\neq 0 \wedge n\in\mathbb{Z}^- \wedge m\in\mathbb{Q}$

1: $\int x^m (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx$ when $b^2-4ac\neq 0 \wedge n\in\mathbb{Z}^- \wedge m\in\mathbb{Z}$

Derivation: Integration by substitution

- **Basis:** $F[x] = -\text{Subst}\left[\frac{F[x^{-1}]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Rule 1.2.3.4.6.2.1.1: If $b^2-4ac\neq 0 \wedge n\in\mathbb{Z}^- \wedge m\in\mathbb{Z}$**, then**

$$\int x^m (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx \longrightarrow -\text{Subst}\left[\int \frac{(d+e x^{-n})^q \left(a+b x^{-n}+c x^{-2n}\right)^p}{x^{m+2}} dx, x, \frac{1}{x}\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  -Subst[Int[(d+e*x^(-n))^q*(a+b*x^(-n)+c*x^(-2*n))^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,d,e,p,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && ILtQ[n,0] && IntegerQ[m]
```

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_,x_Symbol] :=
  -Subst[Int[(d+e*x^(-n))^q*(a+c*x^(-2*n))^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,c,d,e,p,q},x] && EqQ[n2,2*n] && ILtQ[n,0] && IntegerQ[m]
```

2: $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^-\,\wedge\,m\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $n\in\mathbb{Z}\,\wedge\,g>1$**, then** $(f\,x)^m\,F[x^n]=-\frac{g}{f}\,\text{Subst}\left[\frac{F[f^{-n}\,x^{-g\,n}]}{x^{g\,(m+1)+1}},\,x,\,\frac{1}{(f\,x)^{1/g}}\right]\partial_x\frac{1}{(f\,x)^{1/g}}$

Rule 1.2.3.4.6.2.1.2: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^-\,\wedge\,m\in\mathbb{F}$**, let** $g=\text{Denominator}[m]$**, then**

$$\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\longrightarrow\;-\frac{g}{f}\,\text{Subst}\left[\int\frac{(d+e\,f^{-n}\,x^{-g\,n})^q\,\left(a+b\,f^{-n}\,x^{-g\,n}+c\,f^{-2\,n}\,x^{-2\,g\,n}\right)^p}{x^{g\,(m+1)+1}}\,dx,\,x,\,\frac{1}{(f\,x)^{1/g}}\right]$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{g=Denominator[m]},
  -g/f*Subst[Int[(d+e*f^(-n)*x^(-g*n))^q*(a+b*f^(-n)*x^(-g*n)+c*f^(-2*n)*x^(-2*g*n))^p/x^(g*(m+1)+1),x],x,1/(f*x)^(1/g)]] /;
FreeQ[{a,b,c,d,e,f,p,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && ILtQ[n,0] && FractionQ[m]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{g=Denominator[m]},
  -g/f*Subst[Int[(d+e*f^(-n)*x^(-g*n))^q*(a+c*f^(-2*n)*x^(-2*g*n))^p/x^(g*(m+1)+1),x],x,1/(f*x)^(1/g)]] /;
FreeQ[{a,c,d,e,f,p,q},x] && EqQ[n2,2*n] && ILtQ[n,0] && FractionQ[m]
```

2: $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^-\,\wedge\,m\notin\mathbb{Q}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x\left((f\,x)^m\,\left(x^{-1}\right)^m\right)=0$

Basis: $(f\,x)^m\,\left(x^{-1}\right)^m=f^{\text{IntPart}[m]}\,(f\,x)^{\text{FracPart}[m]}\,\left(x^{-1}\right)^{\text{FracPart}[m]}$

Basis: $F[x]=-\text{Subst}\left[\frac{F[x^{-1}]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Rule 1.2.3.4.6.2.2: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^-\,\wedge\,m\notin\mathbb{Q}$**, then**

$$\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\longrightarrow\;f^{\text{IntPart}[m]}\,(f\,x)^{\text{FracPart}[m]}\,\left(x^{-1}\right)^{\text{FracPart}[m]}\int\frac{(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{\left(x^{-1}\right)^m}\,dx$$

$$\longrightarrow -f^{\text{IntPart}[m]}\,(f\,x)^{\text{FracPart}[m]}\,\left(x^{-1}\right)^{\text{FracPart}[m]}\,\text{Subst}\left[\int\frac{\left(d+e\,x^{-n}\right)^q\,\left(a+b\,x^{-n}+c\,x^{-2\,n}\right)^p}{x^{m+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  -f^IntPart[m]*(f*x)^FracPart[m]*(x^(-1))^FracPart[m]*Subst[Int[(d+e*x^(-n))^q*(a+b*x^(-n)+c*x^(-2*n))^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,d,e,f,m,p,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && ILtQ[n,0] && Not[RationalQ[m]]
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_,x_Symbol] :=
  -f^IntPart[m]*(f*x)^FracPart[m]*(x^(-1))^FracPart[m]*Subst[Int[(d+e*x^(-n))^q*(a+c*x^(-2*n))^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,c,d,e,f,m,p,q},x] && EqQ[n2,2*n] && ILtQ[n,0] && Not[RationalQ[m]]
```

7. $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{F}$

1: $\int x^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $g\in\mathbb{Z}^+$, **then** $x^m\,F[x^n]=g\,\text{Subst}\left[x^{g\,(m+1)-1}\,F[x^{g\,n}],\,x,\,x^{1/g}\right]\,\partial_x x^{1/g}$

Rule 1.2.3.4.7.1: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{F}$, **let** $g=\text{Denominator}[n]$, **then**

$$\int x^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\longrightarrow g\,\text{Subst}\left[\int x^{g\,(m+1)-1}\,\left(d+e\,x^{g\,n}\right)^q\,\left(a+b\,x^{g\,n}+c\,x^{2\,g\,n}\right)^p\,dx,\,x,\,x^{1/g}\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{g=Denominator[n]},
  g*Subst[Int[x^(g*(m+1)-1)*(d+e*x^(g*n))^q*(a+b*x^(g*n)+c*x^(2*g*n))^p,x],x,x^(1/g)]] /;
FreeQ[{a,b,c,d,e,m,p,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && FractionQ[n]
```

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{g=Denominator[n]},
  g*Subst[Int[x^(g*(m+1)-1)*(d+e*x^(g*n))^q*(a+c*x^(2*g*n))^p,x],x,x^(1/g)]] /;
FreeQ[{a,c,d,e,m,p,q},x] && EqQ[n2,2*n] && FractionQ[n]
```

2: $\int (f x)^m (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx$ when $b^2-4ac\neq 0 \wedge n\in\mathbb{F}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(f x)^m}{x^m} = 0$
- **Basis:** $\frac{(f x)^m}{x^m} = \frac{f^{\text{IntPart}[m]} (f x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule 1.2.3.4.7.2: If $b^2-4ac\neq 0 \wedge n\in\mathbb{F}$**, then**

$$\int (f x)^m (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx \longrightarrow \frac{f^{\text{IntPart}[m]} (f x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx$$

Program code:

```
Int[(f_*x_)^m_*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  f^IntPart[m]*(f*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(d+e*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,f,m,p,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && FractionQ[n]
```

```
Int[(f_*x_)^m_*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_,x_Symbol] :=
  f^IntPart[m]*(f*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(d+e*x^n)^q*(a+c*x^(2*n))^p,x] /;
FreeQ[{a,c,d,e,f,m,p,q},x] && EqQ[n2,2*n] && FractionQ[n]
```

8. $\int (f x)^m (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx$ when $b^2-4ac\neq 0 \wedge \frac{n}{m+1}\in\mathbb{Z}$

1: $\int x^m (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx$ when $b^2-4ac\neq 0 \wedge \frac{n}{m+1}\in\mathbb{Z}$

Derivation: Integration by substitution

- **Basis: If** $\frac{n}{m+1}\in\mathbb{Z}$**, then** $x^m F[x^n] = \frac{1}{m+1} \text{Subst}\left[F\left[x^{\frac{n}{m+1}}\right], x, x^{m+1}\right] \partial_x x^{m+1}$
- **Rule 1.2.3.4.8.1: If** $b^2-4ac\neq 0 \wedge \frac{n}{m+1}\in\mathbb{Z}$

$$\int x^m (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx \longrightarrow \frac{1}{m+1} \text{Subst}\left[\int \left(d+e x^{\frac{n}{m+1}}\right)^q \left(a+b x^{\frac{n}{m+1}}+c x^{\frac{2n}{m+1}}\right)^p dx, x, x^{m+1}\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  1/(m+1)*Subst[Int[(d+e*x^Simplify[n/(m+1)])^q*(a+b*x^Simplify[n/(m+1)]+c*x^Simplify[2*n/(m+1)])^p,x],x,x^(m+1)] /;
FreeQ[{a,b,c,d,e,m,n,p,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_,x_Symbol] :=
  1/(m+1)*Subst[Int[(d+e*x^Simplify[n/(m+1)])^q*(a+c*x^Simplify[2*n/(m+1)])^p,x],x,x^(m+1)] /;
FreeQ[{a,c,d,e,m,n,p,q},x] && EqQ[n2,2*n] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

2: $\int (f\,x)^m \left(d+e\,x^n\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \bigwedge\ \frac{n}{m+1}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(f\,x)^m}{x^m} == 0$
- **Basis:** $\frac{(f\,x)^m}{x^m} == \frac{f^{\text{IntPart}[m]}\,(f\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$
- **Rule 1.2.3.4.8.2: If** $b^2-4\,a\,c\neq 0\ \bigwedge\ \frac{n}{m+1}\in\mathbb{Z}$**, then**

$$\int (f\,x)^m \left(d+e\,x^n\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx \longrightarrow \frac{f^{\text{IntPart}[m]}\,(f\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m \left(d+e\,x^n\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[(f_*x_)^m_*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  f^IntPart[m]*(f*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(d+e*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,f,m,p,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

```
Int[(f_*x_)^m_*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_,x_Symbol] :=
  f^IntPart[m]*(f*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(d+e*x^n)^q*(a+c*x^(2*n))^p,x] /;
FreeQ[{a,c,d,e,f,m,p,q},x] && EqQ[n2,2*n] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

9: $\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0$

Derivation: Algebraic expansion

- **Basis: If** $r=\sqrt{b^2-4\,a\,c}$ **, then** $\frac{1}{a+b\,z+c\,z^2}==\frac{2\,c}{r\,(b-r+2\,c\,z)}-\frac{2\,c}{r\,(b+r+2\,c\,z)}$

Rule 1.2.3.4.9: If $b^2-4\,a\,c\neq 0$**, then**

$$\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx\;\longrightarrow\;\frac{2\,c}{r}\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{b-r+2\,c\,x^n}\,dx-\frac{2\,c}{r}\int \frac{(f\,x)^m\,(d+e\,x^n)^q}{b+r+2\,c\,x^n}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_/(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  With[{r=Rt[b^2-4*a*c,2]},
  2*c/r*Int[(f*x)^m*(d+e*x^n)^q/(b-r+2*c*x^n),x] - 2*c/r*Int[(f*x)^m*(d+e*x^n)^q/(b+r+2*c*x^n),x]] /;
FreeQ[{a,b,c,d,e,f,m,n,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_/(a_+c_.*x_^n2_.),x_Symbol] :=
  With[{r=Rt[-a*c,2]},
  -c/(2*r)*Int[(f*x)^m*(d+e*x^n)^q/(r-c*x^n),x] - c/(2*r)*Int[(f*x)^m*(d+e*x^n)^q/(r+c*x^n),x]] /;
FreeQ[{a,c,d,e,f,m,n,q},x] && EqQ[n2,2*n]
```

10: $\int (f\,x)^m\,(d+e\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,p+1\in\mathbb{Z}^-$

Derivation: Trinomial recurrence 2b

Rule 1.2.3.4.10: If $b^2-4\,a\,c\neq 0\,\wedge\,p+1\in\mathbb{Z}^-$, then

$$\int (f\,x)^m\,(d+e\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow$$
$$-\left((f\,x)^{m+1}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}\,\left(d\,\left(b^2-2\,a\,c\right)-a\,b\,e+(b\,d-2\,a\,e)\,c\,x^n\right)\right)\Big/\left(a\,f\,n\,(p+1)\,\left(b^2-4\,a\,c\right)\right)+\frac{1}{a\,n\,(p+1)\,\left(b^2-4\,a\,c\right)}\int (f\,x)^m\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}\cdot$$
$$\left(d\,\left(b^2\,(m+n\,(p+1)+1)-2\,a\,c\,(m+2\,n\,(p+1)+1)\right)-a\,b\,e\,(m+1)+(m+n\,(2\,p+3)+1)\,(b\,d-2\,a\,e)\,c\,x^n\right)\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  -(f*x)^(m+1)*(a+b*x^n+c*x^(2*n))^(p+1)*(d*(b^2-2*a*c)-a*b*e+(b*d-2*a*e)*c*x^n)/(a*f*n*(p+1)*(b^2-4*a*c)) +
  1/(a*n*(p+1)*(b^2-4*a*c))*Int[(f*x)^m*(a+b*x^n+c*x^(2*n))^(p+1)*
       Simp[d*(b^2*(m+n*(p+1)+1)-2*a*c*(m+2*n*(p+1)+1))-a*b*e*(m+1)+(m+n*(2*p+3)+1)*(b*d-2*a*e)*c*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && ILtQ[p+1,0]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)*(a_+c_.*x_^n2_)^p_,x_Symbol] :=
  -(f*x)^(m+1)*(a+c*x^(2*n))^(p+1)*(d+e*x^n)/(2*a*f*n*(p+1)) +
  1/(2*a*n*(p+1))*Int[(f*x)^m*(a+c*x^(2*n))^(p+1)*Simp[d*(m+2*n*(p+1)+1)+e*(m+n*(2*p+3)+1)*x^n,x],x] /;
FreeQ[{a,c,d,e,f,m,n},x] && EqQ[n2,2*n] && ILtQ[p+1,0]
```

11: $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,(p\in\mathbb{Z}^+\vee q\in\mathbb{Z}^+)$

Derivation: Algebraic expansion

Rule 1.2.3.4.11: If $b^2-4\,a\,c\neq 0\,\wedge\,(p\in\mathbb{Z}^+\vee q\in\mathbb{Z}^+)$, then

$$\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow\,\int \text{ExpandIntegrand}\left[(f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p,\,x\right]dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m*(d+e*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x],x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && (IGtQ[p,0] || IGtQ[q,0])
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m*(d+e*x^n)^q*(a+c*x^(2*n))^p,x],x] /;
FreeQ[{a,c,d,e,f,m,n,p,q},x] && EqQ[n2,2*n] && (IGtQ[p,0] || IGtQ[q,0])
```

12: $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+c\,x^{2\,n}\right)^p\,dx$ when $p\notin\mathbb{Z}\,\land\,q\in\mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: If $q\in\mathbb{Z}$, **then** $(d+e\,x^n)^q = \left(\frac{d}{d^2-e^2\,x^{2\,n}}-\frac{e\,x^n}{d^2-e^2\,x^{2\,n}}\right)^{-q}$

Note: Resulting integrands are of the form $x^m\,\left(a+b\,x^{2\,n}\right)^p\,\left(c+d\,x^{2\,n}\right)^q$ **which are integrable in terms of the Appell hypergeometric function .**

Rule 1.2.3.4.12: If $p\notin\mathbb{Z}\,\land\,q\in\mathbb{Z}^-$, **then**

$$\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow\,\frac{(f\,x)^m}{x^m}\int x^m\,\left(a+c\,x^{2\,n}\right)^p\,\text{ExpandIntegrand}\left[\left(\frac{d}{d^2-e^2\,x^{2\,n}}-\frac{e\,x^n}{d^2-e^2\,x^{2\,n}}\right)^{-q},\,x\right]dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_*(a_+c_.*x_^n2_)^p_,x_Symbol] :=
  (f*x)^m/x^m*Int[ExpandIntegrand[x^m*(a+c*x^(2*n))^p,(d/(d^2-e^2*x^(2*n))-e*x^n/(d^2-e^2*x^(2*n)))^(-q),x],x] /;
FreeQ[{a,c,d,e,f,m,n,p},x] && EqQ[n2,2*n] && Not[IntegerQ[p]] && ILtQ[q,0]
```

U: $\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$

Rule 1.2.3.4.X:

$$\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow\,\int (f\,x)^m\,(d+e\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Unintegrable[(f*x)^m*(d+e*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q},x] && EqQ[n2,2*n]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Unintegrable[(f*x)^m*(d+e*x^n)^q*(a+c*x^(2*n))^p,x] /;
FreeQ[{a,c,d,e,f,m,n,p,q},x] && EqQ[n2,2*n]
```

S: $\int u^m (d+e\,v^n)^q \left(a+b\,v^n+c\,v^{2\,n}\right)^p dx$ when $v == f+g\,x \wedge u == h\,v$

Derivation: Integration by substitution and piecewise constant extraction

Basis: If $u == h\,v$**, then** $\partial_x \frac{u^m}{v^m} == 0$

Rule 1.2.3.4.S: If $v == f+g\,x \wedge u == h\,v$**, then**

$$\int u^m (d+e\,v^n)^q \left(a+b\,v^n+c\,v^{2\,n}\right)^p dx \longrightarrow \frac{u^m}{g\,v^m}\,\text{Subst}\left[\int x^m (d+e\,x^n)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx,\,x,\,v\right]$$

Program code:

```
Int[u_^m_.*(d_+e_.*v_^n_)^q_.*(a_+b_.*v_^n_+c_.*v_^n2_.)^p_.,x_Symbol] :=
  u^m/(Coefficient[v,x,1]*v^m)*Subst[Int[x^m*(d+e*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x],x,v] /;
FreeQ[{a,b,c,d,e,m,n,p,q},x] && EqQ[n2,2*n] && LinearPairQ[u,v,x] && NeQ[v,x]
```

```
Int[u_^m_.*(d_+e_.*v_^n_)^q_.*(a_+c_.*v_^n2_.)^p_.,x_Symbol] :=
  u^m/(Coefficient[v,x,1]*v^m)*Subst[Int[x^m*(d+e*x^n)^q*(a+c*x^(2*n))^p,x],x,v] /;
FreeQ[{a,c,d,e,m,n,p},x] && EqQ[n2,2*n] && LinearPairQ[u,v,x] && NeQ[v,x]
```

Rules for integrands of the form $(f\,x)^m\,(d+e\,x^{-n})^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p$

1. $\int x^m\,(d+e\,x^{-n})^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $p\in\mathbb{Z}\ \lor\ q\in\mathbb{Z}$

1: $\int x^m\,(d+e\,x^{-n})^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $q\in\mathbb{Z}\ \land\ (n>0\ \lor\ p\notin\mathbb{Z})$

Derivation: Algebraic simplification

Basis: If $q\in\mathbb{Z}$**, then** $(d+e\,x^{-n})^q = x^{-n\,q}\,(e+d\,x^n)^q$

Rule: If $q\in\mathbb{Z}\ \land\ (n>0\ \lor\ p\notin\mathbb{Z})$**, then**

$$\int x^m\,(d+e\,x^{-n})^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\ \longrightarrow\ \int x^{m-n\,q}\,(e+d\,x^n)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^mn_.)^q_.*(a_.+b_.*x_^n_.+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[x^(m-n*q)*(e+d*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && EqQ[mn,-n] && IntegerQ[q] && (PosQ[n] || Not[IntegerQ[p]])
```

```
Int[x_^m_.*(d_+e_.*x_^mn_.)^q_.*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[x^(m+mn*q)*(e+d*x^(-mn))^q*(a+c*x^n2)^p,x] /;
FreeQ[{a,c,d,e,m,mn,p},x] && EqQ[n2,-2*mn] && IntegerQ[q] && (PosQ[n2] || Not[IntegerQ[p]])
```

2: $\int x^m\,(d+e\,x^n)^q\,\left(a+b\,x^{-n}+c\,x^{-2\,n}\right)^p\,dx$ when $p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $p\in\mathbb{Z}$**, then** $\left(a+b\,x^{-n}+c\,x^{-2\,n}\right)^p = x^{-2\,n\,p}\,\left(c+b\,x^n+a\,x^{2\,n}\right)^p$

Rule: If $p\in\mathbb{Z}$**, then**

$$\int x^m\,(d+e\,x^n)^q\,\left(a+b\,x^{-n}+c\,x^{-2\,n}\right)^p\,dx\ \longrightarrow\ \int x^{m-2\,n\,p}\,(d+e\,x^n)^q\,\left(c+b\,x^n+a\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^n_.)^q_.*(a_.+b_.*x_^mn_.+c_.*x_^mn2_.)^p_.,x_Symbol] :=
  Int[x^(m-2*n*p)*(d+e*x^n)^q*(c+b*x^n+a*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,m,n,q},x] && EqQ[mn,-n] && EqQ[mn2,2*mn] && IntegerQ[p]
```

```
Int[x_^m_.*(d_+e_.*x_^n_.)^q_.*(a_.+c_.*x_^mn2_.)^p_.,x_Symbol] :=
  Int[x^(m-2*n*p)*(d+e*x^n)^q*(c+a*x^(2*n))^p,x] /;
FreeQ[{a,c,d,e,m,n,q},x] && EqQ[mn2,-2*n] && IntegerQ[p]
```

2. $\int x^m \left(d+e\,x^{-n}\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $p\notin\mathbb{Z}\,\land\,q\notin\mathbb{Z}$

1: $\int x^m \left(d+e\,x^{-n}\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $p\notin\mathbb{Z}\,\land\,q\notin\mathbb{Z}\,\land\,n>0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^{n\,q}\left(d+e\,x^{-n}\right)^q}{\left(1+\frac{d\,x^n}{e}\right)^q}=0$

Rule: If $p\notin\mathbb{Z}\,\land\,q\notin\mathbb{Z}\,\land\,n>0$**, then**

$$\int x^m \left(d+e\,x^{-n}\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx\;\longrightarrow\;\frac{e^{\text{IntPart}[q]}\,x^{n\,\text{FracPart}[q]}\left(d+e\,x^{-n}\right)^{\text{FracPart}[q]}}{\left(1+\frac{d\,x^n}{e}\right)^{\text{FracPart}[q]}}\int x^{m-n\,q}\left(1+\frac{d\,x^n}{e}\right)^q\left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^mn_.)^q_*(a_.+b_.*x_^n_.+c_.*x_^n2_.)^p_.,x_Symbol] :=
  e^IntPart[q]*x^(n*FracPart[q])*(d+e*x^(-n))^FracPart[q]/(1+d*x^n/e)^FracPart[q]*Int[x^(m-n*q)*(1+d*x^n/e)^q*(a+b*x^n+c*x^(2*n))
FreeQ[{a,b,c,d,e,m,n,p,q},x] && EqQ[n2,2*n] && EqQ[mn,-n] && Not[IntegerQ[p]] && Not[IntegerQ[q]] && PosQ[n]
```

```
Int[x_^m_.*(d_+e_.*x_^mn_.)^q_*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  e^IntPart[q]*x^(-mn*FracPart[q])*(d+e*x^mn)^FracPart[q]/(1+d*x^(-mn)/e)^FracPart[q]*Int[x^(m+mn*q)*(1+d*x^(-mn)/e)^q*(a+c*x^n2
FreeQ[{a,c,d,e,m,mn,p,q},x] && EqQ[n2,-2*mn] && Not[IntegerQ[p]] && Not[IntegerQ[q]] && PosQ[n2]
```

x: $\int x^m \left(d+e\,x^{-n}\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $p \notin \mathbb{Z} \land q \notin \mathbb{Z} \land n>0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{x^{n\,q}\,(d+e\,x^{-n})^q}{(e+d\,x^n)^q} == 0$

Rule: If $p \notin \mathbb{Z} \land q \notin \mathbb{Z} \land n>0$, **then**

$$\int x^m \left(d+e\,x^{-n}\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx \to \frac{x^{n\,\text{FracPart}[q]} \left(d+e\,x^{-n}\right)^{\text{FracPart}[q]}}{\left(e+d\,x^n\right)^{\text{FracPart}[q]}} \int x^{m-n\,q} \left(e+d\,x^n\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
(* Int[x_^m_.*(d_+e_.*x_^mn_.)^q_*(a_.+b_.*x_^n_.+c_.*x_^n2_.)^p_.,x_Symbol] :=
  x^(n*FracPart[q])*(d+e*x^(-n))^FracPart[q]/(e+d*x^n)^FracPart[q]*Int[x^(m-n*q)*(e+d*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p,q},x] && EqQ[n2,2*n] && EqQ[mn,-n] && Not[IntegerQ[p]] && Not[IntegerQ[q]] && PosQ[n] *)
```

```
(* Int[x_^m_.*(d_+e_.*x_^mn_.)^q_*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  x^(-mn*FracPart[q])*(d+e*x^mn)^FracPart[q]/(e+d*x^(-mn))^FracPart[q]*Int[x^(m+mn*q)*(e+d*x^(-mn))^q*(a+c*x^n2)^p,x] /;
FreeQ[{a,c,d,e,m,mn,p,q},x] && EqQ[n2,-2*mn] && Not[IntegerQ[p]] && Not[IntegerQ[q]] && PosQ[n2] *)
```

2: $\int x^m \left(d+e\,x^n\right)^q \left(a+b\,x^{-n}+c\,x^{-2\,n}\right)^p\,dx$ when $p \notin \mathbb{Z} \land q \notin \mathbb{Z} \land n>0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{x^{2\,n\,p}\,(a+b\,x^{-n}+c\,x^{-2\,n})^p}{(c+b\,x^n+a\,x^{2\,n})^p} == 0$

Rule: If $p \notin \mathbb{Z} \land q \notin \mathbb{Z} \land n>0$, **then**

$$\int x^m \left(d+e\,x^n\right)^q \left(a+b\,x^{-n}+c\,x^{-2\,n}\right)^p\,dx \to \frac{x^{2\,n\,\text{FracPart}[p]} \left(a+b\,x^{-n}+c\,x^{-2\,n}\right)^{\text{FracPart}[p]}}{\left(c+b\,x^n+a\,x^{2\,n}\right)^{\text{FracPart}[p]}} \int x^{m-2\,n\,p} \left(d+e\,x^n\right)^q \left(c+b\,x^n+a\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^n_.)^q_.*(a_.+b_.*x_^mn_.+c_.*x_^mn2_.)^p_,x_Symbol] :=
  x^(2*n*FracPart[p])*(a+b*x^(-n)+c*x^(-2*n))^FracPart[p]/(c+b*x^n+a*x^(2*n))^FracPart[p]*
    Int[x^(m-2*n*p)*(d+e*x^n)^q*(c+b*x^n+a*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p,q},x] && EqQ[mn,-n] && EqQ[mn2,2*mn] && Not[IntegerQ[p]] && Not[IntegerQ[q]] && PosQ[n]
```

```
Int[x_^m_.*(d_+e_.*x_^n_.)^q_.*(a_.+c_.*x_^mn2_.)^p_,x_Symbol] :=
  x^(2*n*FracPart[p])*(a+c*x^(-2*n))^FracPart[p]/(c+a*x^(2*n))^FracPart[p]*
    Int[x^(m-2*n*p)*(d+e*x^n)^q*(c+a*x^(2*n))^p,x] /;
FreeQ[{a,c,d,e,m,n,p,q},x] && EqQ[mn2,-2*n] && Not[IntegerQ[p]] && Not[IntegerQ[q]] && PosQ[n]
```

3: $\int (f\,x)^m\,(d+e\,x^{-n})^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(f\,x)^m}{x^m} == 0$

Rule:

$$\int (f\,x)^m\,(d+e\,x^{-n})^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx \longrightarrow \frac{f^{\text{IntPart}[m]}\,(f\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,(d+e\,x^{-n})^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[(f_*x_)^m_*(d_+e_.*x_^mn_.)^q_.*(a_.+b_.*x_^n_.+c_.*x_^n2_.)^p_.,x_Symbol] :=
  f^IntPart[m]*(f*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(d+e*x^mn)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q},x] && EqQ[n2,2*n] && EqQ[mn,-n]
```

```
Int[(f_*x_)^m_*(d_+e_.*x_^mn_.)^q_.*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  f^IntPart[m]*(f*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(d+e*x^mn)^q*(a+c*x^n2)^p,x] /;
FreeQ[{a,c,d,e,f,m,mn,p,q},x] && EqQ[n2,-2*mn]
```

Rules for integrands of the form $(f\,x)^m\,(d+e\,x^n)^q\,(a+b\,x^{-n}+c\,x^n)^p$

1. $\int x^m\,(d+e\,x^n)^q\,(a+b\,x^{-n}+c\,x^n)^p\,dx$

1: $\int x^m\,(d+e\,x^n)^q\,(a+b\,x^{-n}+c\,x^n)^p\,dx$ when $p\in\mathbb{Z}$

Derivation: Algebraic normalization

Basis: $a+b\,x^{-n}+c\,x^n == x^{-n}\left(b+a\,x^n+c\,x^{2\,n}\right)$

Rule 1.2.3.4.13.1.1: If $p\in\mathbb{Z}$**, then**

$$\int x^m\,(d+e\,x^n)^q\,(a+b\,x^{-n}+c\,x^n)^p\,dx \longrightarrow \int x^{m-n\,p}\,(d+e\,x^n)^q\,\left(b+a\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^mn_+c_.*x_^n_.)^p_.,x_Symbol] :=
  Int[x^(m-n*p)*(d+e*x^n)^q*(b+a*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,m,n,q},x] && EqQ[mn,-n] && IntegerQ[p]
```

2: $\int x^m\,(d+e\,x^n)^q\,(a+b\,x^{-n}+c\,x^n)^p\,dx$ when $p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{x^{n\,p}\,(a+b\,x^{-n}+c\,x^n)^p}{(b+a\,x^n+c\,x^{2\,n})^p} == 0$
- **Basis:** $\frac{x^{n\,p}\,(a+b\,x^{-n}+c\,x^n)^p}{(b+a\,x^n+c\,x^{2\,n})^p} == \frac{x^{n\,\text{FracPart}[p]}\,(a+b\,x^{-n}+c\,x^n)^{\text{FracPart}[p]}}{(b+a\,x^n+c\,x^{2\,n})^{\text{FracPart}[p]}}$

Rule 1.2.3.4.13.1.2: If $p\notin\mathbb{Z}$**, then**

$$\int x^m\,(d+e\,x^n)^q\,(a+b\,x^{-n}+c\,x^n)^p\,dx \longrightarrow \frac{x^{n\,\text{FracPart}[p]}\,(a+b\,x^{-n}+c\,x^n)^{\text{FracPart}[p]}}{\left(b+a\,x^n+c\,x^{2\,n}\right)^{\text{FracPart}[p]}}\int x^{m-n\,p}\,(d+e\,x^n)^q\,\left(b+a\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^mn_+c_.*x_^n_.)^p_.,x_Symbol] :=
  x^(n*FracPart[p])*(a+b/x^n+c*x^n)^FracPart[p]/(b+a*x^n+c*x^(2*n))^FracPart[p]*
    Int[x^(m-n*p)*(d+e*x^n)^q*(b+a*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p,q},x] && EqQ[mn,-n] && Not[IntegerQ[p]]
```

2: $\int (f\,x)^m\,(d+e\,x^n)^q\,(a+b\,x^{-n}+c\,x^n)^p\,dx$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(f\,x)^m}{x^m} == 0$

Basis: $\frac{(f\,x)^m}{x^m} == \frac{f^{\text{IntPart}[m]}\,(f\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule 1.2.3.4.13.2:

$$\int (f\,x)^m\,(d+e\,x^n)^q\,(a+b\,x^{-n}+c\,x^n)^p\,dx \longrightarrow \frac{f^{\text{IntPart}[m]}\,(f\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,(d+e\,x^n)^q\,(a+b\,x^{-n}+c\,x^n)^p\,dx$$

Program code:

```
Int[(f_*x_)^m_.*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^mn_+c_.*x_^n_.)^p_.,x_Symbol] :=
  f^IntPart[m]*(f*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(d+e*x^n)^q*(a+b*x^(-n)+c*x^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q},x] && EqQ[mn,-n]
```

Rules for integrands of the form $(f\,x)^m\left(d_1+e_1\,x^{n/2}\right)^q\left(d_2+e_2\,x^{n/2}\right)^q\left(a+b\,x^n+c\,x^{2\,n}\right)^p$

1. $\int (f\,x)^m\left(d_1+e_1\,x^{n/2}\right)^q\left(d_2+e_2\,x^{n/2}\right)^q\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $d_2\,e_1+d_1\,e_2 == 0$

1: $\int (f\,x)^m\left(d_1+e_1\,x^{n/2}\right)^q\left(d_2+e_2\,x^{n/2}\right)^q\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $d_2\,e_1+d_1\,e_2 == 0\;\bigwedge\;(q\in\mathbb{Z}\;\bigvee\;d_1>0\;\bigwedge\;d_2>0)$

Derivation: Algebraic simplification

Basis: If $d_2\,e_1+d_1\,e_2 == 0\;\bigwedge\;(q\in\mathbb{Z}\;\bigvee\;d_1>0\wedge d_2>0)$**, then** $\left(d_1+e_1\,x^{n/2}\right)^q\left(d_2+e_2\,x^{n/2}\right)^q == (d_1\,d_2+e_1\,e_2\,x^n)^q$

Rule: If $d_2\,e_1+d_1\,e_2 == 0\;\bigwedge\;(q\in\mathbb{Z}\;\bigvee\;d_1>0\wedge d_2>0)$**, then**

$$\int (f\,x)^m\left(d_1+e_1\,x^{n/2}\right)^q\left(d_2+e_2\,x^{n/2}\right)^q\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx \longrightarrow \int (f\,x)^m\,(d_1\,d_2+e_1\,e_2\,x^n)^q\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d1_+e1_.*x_^non2_.)^q_.*(d2_+e2_.*x_^non2_.)^q_.*(a_.+b_.*x_^n_+c_.*x_^n2_)^p_.,x_Symbol] :=
  Int[(f*x)^m*(d1*d2+e1*e2*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,n,p,q},x] && EqQ[n2,2*n] && EqQ[non2,n/2] && EqQ[d2*e1+d1*e2,0] && (IntegerQ[q] || GtQ[d1,0] && GtQ[d2
```

2: $\int (f\,x)^m \left(d_1+e_1\,x^{n/2}\right)^q \left(d_2+e_2\,x^{n/2}\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $d_2\,e_1+d_1\,e_2==0$

- **Derivation: Piecewise constant extraction**

- **Basis: If $d_2\,e_1+d_1\,e_2==0$, then $\partial_x \frac{\left(d_1+e_1\,x^{n/2}\right)^q \left(d_2+e_2\,x^{n/2}\right)^q}{\left(d_1\,d_2+e_1\,e_2\,x^n\right)^q}==0$**

- **Rule: If $d_2\,e_1+d_1\,e_2==0$, then**

$$\int (f\,x)^m \left(d_1+e_1\,x^{n/2}\right)^q \left(d_2+e_2\,x^{n/2}\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx \longrightarrow$$

$$\frac{\left(d_1+e_1\,x^{n/2}\right)^{\text{FracPart}[q]} \left(d_2+e_2\,x^{n/2}\right)^{\text{FracPart}[q]}}{\left(d_1\,d_2+e_1\,e_2\,x^n\right)^{\text{FracPart}[q]}} \int (f\,x)^m \left(d_1\,d_2+e_1\,e_2\,x^n\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

- **Program code:**

```
Int[(f_.*x_)^m_.*(d1_+e1_.*x_^non2_.)^q_.*(d2_+e2_.*x_^non2_.)^q_.*(a_.+b_.*x_^n_+c_.*x_^n2_)^p_.,x_Symbol] :=
  (d1+e1*x^(n/2))^FracPart[q]*(d2+e2*x^(n/2))^FracPart[q]/(d1*d2+e1*e2*x^n)^FracPart[q]*
    Int[(f*x)^m*(d1*d2+e1*e2*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,n,p,q},x] && EqQ[n2,2*n] && EqQ[non2,n/2] && EqQ[d2*e1+d1*e2,0]
```