

Rules for integrands involving Fresnel integral functions

1. $\int \text{Fresnels}[a + b x]^n dx$

1: $\int \text{Fresnels}[a + b x] dx$

Derivation: Integration by parts

Basis: $\partial_x \text{Fresnels}[a + b x] = b \sin\left[\frac{\pi}{2} (a + b x)^2\right]$

Rule:

$$\int \text{Fresnels}[a + b x] dx \rightarrow \frac{(a + b x) \text{Fresnels}[a + b x]}{b} - \int (a + b x) \sin\left[\frac{\pi}{2} (a + b x)^2\right] dx \rightarrow \frac{(a + b x) \text{Fresnels}[a + b x]}{b} + \frac{\cos\left[\frac{\pi}{2} (a + b x)^2\right]}{b \pi}$$

Program code:

```
Int[Fresnels[a_+b_.x_],x_Symbol] :=
  (a+b*x)*Fresnels[a+b*x]/b + Cos[Pi/2*(a+b*x)^2]/(b*Pi) /;
FreeQ[{a,b},x]
```

```
Int[FresnelC[a_+b_.x_],x_Symbol] :=
  (a+b*x)*FresnelC[a+b*x]/b - Sin[Pi/2*(a+b*x)^2]/(b*Pi) /;
FreeQ[{a,b},x]
```

2: $\int \text{Fresnels}[a + b x]^2 dx$

Derivation: Integration by parts

Basis: $\partial_x \text{Fresnels}[a + b x]^2 = 2 b \sin\left[\frac{\pi}{2} (a + b x)^2\right] \text{Fresnels}[a + b x]$

Rule:

$$\int \text{Fresnels}[a + b x]^2 dx \rightarrow \frac{(a + b x) \text{Fresnels}[a + b x]^2}{b} - 2 \int (a + b x) \sin\left[\frac{\pi}{2} (a + b x)^2\right] \text{Fresnels}[a + b x] dx$$

Program code:

```
Int[Fresnels[a_+b_.x_]^2,x_Symbol] :=
  (a+b*x)*Fresnels[a+b*x]^2/b -
  2*Int[(a+b*x)*Sin[Pi/2*(a+b*x)^2]*Fresnels[a+b*x],x] /;
FreeQ[{a,b},x]
```

```
Int[FresnelC[a_.+b_.x_]^2,x_Symbol] :=
  (a+b*x)*FresnelC[a+b*x]^2/b -
  2*Int[(a+b*x)*Cos[Pi/2*(a+b*x)^2]*FresnelC[a+b*x],x] /;
FreeQ[{a,b},x]
```

X: $\int \text{FresnelS}[a + b x]^n dx$ when $n \neq 1 \wedge n \neq 2$

Rule: If $n \neq 1 \wedge n \neq 2$, then

$$\int \text{FresnelS}[a + b x]^n dx \rightarrow \int \text{FresnelS}[a + b x]^n dx$$

Program code:

```
Int[FresnelS[a_.+b_.x_]^n_,x_Symbol] :=
  Unintegrable[FresnelS[a+b*x]^n,x] /;
FreeQ[{a,b,n},x] && NeQ[n,1] && NeQ[n,2]
```

```
Int[FresnelC[a_.+b_.x_]^n_,x_Symbol] :=
  Unintegrable[FresnelC[a+b*x]^n,x] /;
FreeQ[{a,b,n},x] && NeQ[n,1] && NeQ[n,2]
```

2. $\int (c + d x)^m \text{FresnelS}[a + b x]^n dx$

1. $\int (c + d x)^m \text{FresnelS}[a + b x] dx$

1. $\int (d x)^m \text{FresnelS}[b x] dx$

1: $\int \frac{\text{FresnelS}[b x]}{x} dx$

Derivation: Algebraic expansion

- **Basis:** $\text{FresnelS}[b x] == \frac{1+i}{4} \text{Erf}\left[\frac{\sqrt{\pi}}{2} (1+i) b x\right] + \frac{1-i}{4} \text{Erf}\left[\frac{\sqrt{\pi}}{2} (1-i) b x\right]$
- **Basis:** $\text{FresnelC}[b x] == \frac{1-i}{4} \text{Erf}\left[\frac{\sqrt{\pi}}{2} (1+i) b x\right] + \frac{1+i}{4} \text{Erf}\left[\frac{\sqrt{\pi}}{2} (1-i) b x\right]$

Rule:

$$\int \frac{\text{FresnelS}[b x]}{x} dx \rightarrow \frac{1+i}{4} \int \frac{\text{Erf}\left[\frac{\sqrt{\pi}}{2} (1+i) b x\right]}{x} dx + \frac{1-i}{4} \int \frac{\text{Erf}\left[\frac{\sqrt{\pi}}{2} (1-i) b x\right]}{x} dx$$

Program code:

```
Int[FresnelS[b_.*x_]/x_,x_Symbol] :=
  (1+I)/4*Int[Erf[Sqrt[Pi]/2*(1+I)*b*x]/x,x] + (1-I)/4*Int[Erf[Sqrt[Pi]/2*(1-I)*b*x]/x,x] /;
FreeQ[b,x]
```

```
Int[FresnelC[b_.*x_]/x_,x_Symbol] :=
  (1-I)/4*Int[Erf[Sqrt[Pi]/2*(1+I)*b*x]/x,x] + (1+I)/4*Int[Erf[Sqrt[Pi]/2*(1-I)*b*x]/x,x] /;
FreeQ[b,x]
```

2: $\int (d x)^m \text{FresnelS}[b x] dx$ when $m \neq -1$

Derivation: Integration by parts

Rule: If $m \neq -1$, then

$$\int (d x)^m \text{FresnelS}[b x] dx \rightarrow \frac{(d x)^{m+1} \text{FresnelS}[b x]}{d (m+1)} - \frac{b}{d (m+1)} \int (d x)^{m+1} \sin\left[\frac{\pi}{2} b^2 x^2\right] dx$$

Program code:

```
Int[(d_.*x_)^m_.*FresnelS[b_.*x_],x_Symbol] :=
  (d*x)^(m+1)*FresnelS[b*x]/(d*(m+1)) - b/(d*(m+1))*Int[(d*x)^(m+1)*Sin[Pi/2*b^2*x^2],x] /;
FreeQ[{b,d,m},x] && NeQ[m,-1]
```

```
Int[(d_.*x_)^m_.*FresnelC[b_.*x_],x_Symbol] :=
  (d*x)^(m+1)*FresnelC[b*x]/(d*(m+1)) - b/(d*(m+1))*Int[(d*x)^(m+1)*Cos[Pi/2*b^2*x^2],x] /;
FreeQ[{b,d,m},x] && NeQ[m,-1]
```

2: $\int (c + d x)^m \text{FresnelS}[a + b x] dx$ when $m \in \mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\partial_x \text{FresnelS}[a + b x] = b \sin\left[\frac{\pi}{2} (a + b x)^2\right]$

Rule: If $m \in \mathbb{Z}^+$, then

$$\int (c + d x)^m \text{Fresnels}[a + b x] dx \rightarrow \frac{(c + d x)^{m+1} \text{Fresnels}[a + b x]}{d (m + 1)} - \frac{b}{d (m + 1)} \int (c + d x)^{m+1} \sin\left[\frac{\pi}{2} (a + b x)^2\right] dx$$

Program code:

```
Int[(c_+d_.*x_)^m_.*Fresnels[a_+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*Fresnels[a+b*x]/(d*(m+1)) -
  b/(d*(m+1))*Int[(c+d*x)^(m+1)*Sin[Pi/2*(a+b*x)^2],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

```
Int[(c_+d_.*x_)^m_.*FresnelC[a_+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*FresnelC[a+b*x]/(d*(m+1)) -
  b/(d*(m+1))*Int[(c+d*x)^(m+1)*Cos[Pi/2*(a+b*x)^2],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

2. $\int (c + d x)^m \text{Fresnels}[a + b x]^2 dx$

1: $\int x^m \text{Fresnels}[b x]^2 dx$ when $m \in \mathbb{Z} \wedge m \neq -1$

Derivation: Integration by parts

Basis: $\partial_x \text{Fresnels}[b x]^2 = 2 b \sin\left[\frac{\pi}{2} b^2 x^2\right] \text{Fresnels}[b x]$

Rule: If $m \in \mathbb{Z} \wedge m \neq -1$, then

$$\int x^m \text{Fresnels}[b x]^2 dx \rightarrow \frac{x^{m+1} \text{Fresnels}[b x]^2}{m+1} - \frac{2 b}{m+1} \int x^{m+1} \sin\left[\frac{\pi}{2} b^2 x^2\right] \text{Fresnels}[b x] dx$$

Program code:

```
Int[x_^m_.*Fresnels[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*Fresnels[b*x]^2/(m+1) -
  2*b/(m+1)*Int[x^(m+1)*Sin[Pi/2*b^2*x^2]*Fresnels[b*x],x] /;
FreeQ[b,x] && IntegerQ[m] && NeQ[m,-1]
```

```
Int[x_^m_.*FresnelC[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*FresnelC[b*x]^2/(m+1) -
  2*b/(m+1)*Int[x^(m+1)*Cos[Pi/2*b^2*x^2]*FresnelC[b*x],x] /;
FreeQ[b,x] && IntegerQ[m] && NeQ[m,-1]
```

2: $\int (c + d x)^m \text{FresnelS}[a + b x]^2 dx$ when $m \in \mathbb{Z}^+$

Derivation: Integration by substitution

Rule: If $m \in \mathbb{Z}^+$, then

$$\int (c + d x)^m \text{FresnelS}[a + b x]^2 dx \rightarrow \frac{1}{b^{m+1}} \text{Subst}\left[\int \text{FresnelS}[x]^2 \text{ExpandIntegrand}[(b c - a d + d x)^m, x] dx, x, a + b x\right]$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*FresnelS[a_.+b_.*x_]^2,x_Symbol] :=
  1/b^(m+1)*Subst[Int[ExpandIntegrand[FresnelS[x]^2,(b*c-a*d+d*x)^m,x],x],x,a+b*x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_.*FresnelC[a_.+b_.*x_]^2,x_Symbol] :=
  1/b^(m+1)*Subst[Int[ExpandIntegrand[FresnelC[x]^2,(b*c-a*d+d*x)^m,x],x],x,a+b*x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

X: $\int (c + d x)^m \text{FresnelS}[a + b x]^n dx$

Rule:

$$\int (c + d x)^m \text{FresnelS}[a + b x]^n dx \rightarrow \int (c + d x)^m \text{FresnelS}[a + b x]^n dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*FresnelS[a_.+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[(c+d*x)^m*FresnelS[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,m,n},x]
```

```
Int[(c_.+d_.*x_)^m_.*FresnelC[a_.+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[(c+d*x)^m*FresnelC[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,m,n},x]
```

3. $\int e^{c+dx^2} \text{Fresnels}[a+bx]^n dx$

1: $\int e^{c+dx^2} \text{Fresnels}[bx] dx$ when $d^2 = -\frac{\pi^2}{4} b^4$

Derivation: Algebraic expansion

■ Basis: $\text{Fresnels}[bx] = \frac{1+i}{4} \text{Erf}\left[\frac{\sqrt{\pi}}{2} (1+i) bx\right] + \frac{1-i}{4} \text{Erf}\left[\frac{\sqrt{\pi}}{2} (1-i) bx\right]$

■ Basis: $\text{FresnelC}[bx] = \frac{1-i}{4} \text{Erf}\left[\frac{\sqrt{\pi}}{2} (1+i) bx\right] + \frac{1+i}{4} \text{Erf}\left[\frac{\sqrt{\pi}}{2} (1-i) bx\right]$

■ Note: If $d^2 = -\frac{\pi^2}{4} b^4$, then resulting integrands are integrable.

Rule:

$$\int e^{c+dx^2} \text{Fresnels}[bx] dx \rightarrow \frac{1+i}{4} \int e^{c+dx^2} \text{Erf}\left[\frac{\sqrt{\pi}}{2} (1+i) bx\right] dx + \frac{1-i}{4} \int e^{c+dx^2} \text{Erf}\left[\frac{\sqrt{\pi}}{2} (1-i) bx\right] dx$$

Program code:

```
Int[E^(c_.+d_.*x_^2)*Fresnels[b_.*x_],x_Symbol] :=
  (1+I)/4*Int[E^(c+d*x^2)*Erf[Sqrt[Pi]/2*(1+I)*b*x],x] + (1-I)/4*Int[E^(c+d*x^2)*Erf[Sqrt[Pi]/2*(1-I)*b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,-Pi^2/4*b^4]
```

```
Int[E^(c_.+d_.*x_^2)*FresnelC[b_.*x_],x_Symbol] :=
  (1-I)/4*Int[E^(c+d*x^2)*Erf[Sqrt[Pi]/2*(1+I)*b*x],x] + (1+I)/4*Int[E^(c+d*x^2)*Erf[Sqrt[Pi]/2*(1-I)*b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,-Pi^2/4*b^4]
```

X: $\int e^{c+dx^2} \text{Fresnels}[a+bx]^n dx$

Rule:

$$\int e^{c+dx^2} \text{Fresnels}[a+bx]^n dx \rightarrow \int e^{c+dx^2} \text{Fresnels}[a+bx]^n dx$$

Program code:

```
Int[E^(c_.+d_.*x_^2)*Fresnels[a_.+b_.*x_]^n_. ,x_Symbol] :=
  Unintegrable[E^(c+d*x^2)*Fresnels[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x]
```

```
Int[E^(c_.+d_.*x_^2)*FresnelC[a_.+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[E^(c+d*x^2)*FresnelC[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x]
```

4. $\int \sin[c + d x^2] \operatorname{FresnelS}[a + b x]^n dx$

1: $\int \sin[d x^2] \operatorname{FresnelS}[b x]^n dx$ when $d^2 = \frac{\pi^2}{4} b^4$

Derivation: Integration by substitution

- **Basis:** If $d^2 = \frac{\pi^2}{4} b^4$, then $\sin[d x^2] F[\operatorname{FresnelS}[b x]] = \frac{\pi b}{2 d} \operatorname{Subst}[F[x], x, \operatorname{FresnelS}[b x]] \partial_x \operatorname{FresnelS}[b x]$
- **Rule:** If $d^2 = \frac{\pi^2}{4} b^4$, then

$$\int \sin[d x^2] \operatorname{FresnelS}[b x]^n dx \rightarrow \frac{\pi b}{2 d} \operatorname{Subst}\left[\int x^n dx, x, \operatorname{FresnelS}[b x]\right]$$

Program code:

```
Int[Sin[d_.*x_^2]*FresnelS[b_.*x_]^n_,x_Symbol] :=
  Pi*b/(2*d)*Subst[Int[x^n,x],x,FresnelS[b*x]] /;
FreeQ[{b,d,n},x] && EqQ[d^2,Pi^2/4*b^4]
```

```
Int[Cos[d_.*x_^2]*FresnelC[b_.*x_]^n_,x_Symbol] :=
  Pi*b/(2*d)*Subst[Int[x^n,x],x,FresnelC[b*x]] /;
FreeQ[{b,d,n},x] && EqQ[d^2,Pi^2/4*b^4]
```

2: $\int \sin[c + d x^2] \text{FresnelS}[b x] dx$ when $d^2 = \frac{\pi^2}{4} b^4$

Derivation: Algebraic expansion

- **Basis:** $\sin[c + d x^2] = \sin[c] \cos[d x^2] + \cos[c] \sin[d x^2]$

- **Rule:** If $d^2 = \frac{\pi^2}{4} b^4$, then

$$\int \sin[c + d x^2] \text{FresnelS}[b x] dx \rightarrow \sin[c] \int \cos[d x^2] \text{FresnelS}[b x] dx + \cos[c] \int \sin[d x^2] \text{FresnelS}[b x] dx$$

Program code:

```
Int[Sin[c_+d_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  Sin[c]*Int[Cos[d*x^2]*FresnelS[b*x],x] + Cos[c]*Int[Sin[d*x^2]*FresnelS[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

```
Int[Cos[c_+d_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  Cos[c]*Int[Cos[d*x^2]*FresnelC[b*x],x] - Sin[c]*Int[Sin[d*x^2]*FresnelC[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

X: $\int \sin[c + d x^2] \text{FresnelS}[a + b x]^n dx$

- **Rule:**

$$\int \sin[c + d x^2] \text{FresnelS}[a + b x]^n dx \rightarrow \int \sin[c + d x^2] \text{FresnelS}[a + b x]^n dx$$

- **Program code:**

```
Int[Sin[c_+d_.*x_^2]*FresnelS[a_+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[Sin[c+d*x^2]*FresnelS[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x]
```

```
Int[Cos[c_+d_.*x_^2]*FresnelC[a_+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[Cos[c+d*x^2]*FresnelC[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x]
```


5. $\int \cos[c + d x^2] \text{Fresnels}[a + b x]^n dx$

1: $\int \cos[d x^2] \text{Fresnels}[b x] dx$ when $d^2 = \frac{\pi^2}{4} b^4$

Derivation: Algebraic expansion

Rule: If $d^2 = \frac{\pi^2}{4} b^4$, then

$$\int \cos[d x^2] \text{Fresnels}[b x] dx \rightarrow$$

$$\frac{\text{FresnelC}[b x] \text{Fresnels}[b x]}{2 b} - \frac{1}{8} i b x^2 \text{HypergeometricPFQ}\left[\{1, 1\}, \left\{\frac{3}{2}, 2\right\}, -\frac{1}{2} i b^2 \pi x^2\right] + \frac{1}{8} i b x^2 \text{HypergeometricPFQ}\left[\{1, 1\}, \left\{\frac{3}{2}, 2\right\}, \frac{1}{2} i b^2 \pi x^2\right]$$

Program code:

```
Int[Cos[d_.**x_^2]*Fresnels[b_.**x_],x_Symbol] :=
  FresnelC[b*x]*Fresnels[b*x]/(2*b) -
  1/8*I*b*x^2*HypergeometricPFQ[{1,1},{3/2,2},-1/2*I*b^2*Pi*x^2] +
  1/8*I*b*x^2*HypergeometricPFQ[{1,1},{3/2,2},1/2*I*b^2*Pi*x^2] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

```
Int[Sin[d_.**x_^2]*FresnelC[b_.**x_],x_Symbol] :=
  b*Pi*FresnelC[b*x]*Fresnels[b*x]/(4*d) +
  1/8*I*b*x^2*HypergeometricPFQ[{1,1},{3/2,2},-I*d*x^2] -
  1/8*I*b*x^2*HypergeometricPFQ[{1,1},{3/2,2},I*d*x^2] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

2: $\int \cos[c + d x^2] \text{FresnelS}[b x] dx$ when $d^2 = \frac{\pi^2}{4} b^4$

Derivation: Algebraic expansion

- **Basis:** $\cos[c + d x^2] = \cos[c] \cos[d x^2] - \sin[c] \sin[d x^2]$

- **Rule:** If $d^2 = \frac{\pi^2}{4} b^4$, then

$$\int \cos[c + d x^2] \text{FresnelS}[b x] dx \rightarrow \cos[c] \int \cos[d x^2] \text{FresnelS}[b x] dx - \sin[c] \int \sin[d x^2] \text{FresnelS}[b x] dx$$

Program code:

```
Int[Cos[c_+d_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  Cos[c]*Int[Cos[d*x^2]*FresnelS[b*x],x] - Sin[c]*Int[Sin[d*x^2]*FresnelS[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

```
Int[Sin[c_+d_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  Sin[c]*Int[Cos[d*x^2]*FresnelC[b*x],x] + Cos[c]*Int[Sin[d*x^2]*FresnelC[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

X: $\int \cos[c + d x^2] \text{FresnelS}[a + b x]^n dx$

- **Rule:**

$$\int \cos[c + d x^2] \text{FresnelS}[a + b x]^n dx \rightarrow \int \cos[c + d x^2] \text{FresnelS}[a + b x]^n dx$$

- **Program code:**

```
Int[Cos[c_+d_.*x_^2]*FresnelS[a_+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[Cos[c+d*x^2]*FresnelS[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x]
```

```
Int[Sin[c_+d_.*x_^2]*FresnelC[a_+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[Sin[c+d*x^2]*FresnelC[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x]
```

$$6. \int (e x)^m \sin[c + d x^2] \operatorname{Fresnels}[a + b x]^n dx$$

$$1. \int x^m \sin[d x^2] \operatorname{Fresnels}[b x] dx \text{ when } d^2 = \frac{\pi^2}{4} b^4 \bigwedge m \in \mathbb{Z}$$

$$1. \int x^m \sin[d x^2] \operatorname{Fresnels}[b x] dx \text{ when } d^2 = \frac{\pi^2}{4} b^4 \bigwedge m \in \mathbb{Z}^+$$

$$\textcolor{red}{1}: \int x \sin[d x^2] \operatorname{Fresnels}[b x] dx \text{ when } d^2 = \frac{\pi^2}{4} b^4$$

Derivation: Integration by parts and algebraic simplification

- **Basis:** $-\partial_x \frac{\cos[d x^2]}{2 d} = x \sin[d x^2]$
- **Basis:** If $d^2 = \frac{\pi^2}{4} b^4$, then $\cos[d x^2] \sin\left[\frac{1}{2} b^2 \pi x^2\right] = \frac{d}{b^2 \pi} \sin[2 d x^2]$
- **Rule:** If $d^2 = \frac{\pi^2}{4} b^4$, then

$$\int x \sin[d x^2] \operatorname{Fresnels}[b x] dx \rightarrow -\frac{\cos[d x^2] \operatorname{Fresnels}[b x]}{2 d} + \frac{1}{2 b \pi} \int \sin[2 d x^2] dx$$

- **Basis:** $\partial_x \frac{\sin[d x^2]}{2 d} = x \cos[d x^2]$
- **Basis:** If $d^2 = \frac{\pi^2}{4} b^4$, then $\sin[d x^2] \cos\left[\frac{1}{2} b^2 \pi x^2\right] = \frac{1}{2} \sin[2 d x^2]$
- **Rule:** If $d^2 = \frac{\pi^2}{4} b^4$, then

$$\int x \cos[d x^2] \operatorname{FresnelC}[b x] dx \rightarrow \frac{\sin[d x^2] \operatorname{FresnelC}[b x]}{2 d} - \frac{b}{4 d} \int \sin[2 d x^2] dx$$

■ **Program code:**

```
Int[x_*Sin[d_*x^2]*Fresnels[b_*x],x_Symbol] :=
  -Cos[d*x^2]*Fresnels[b*x]/(2*d) + 1/(2*b*Pi)*Int[Sin[2*d*x^2],x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

```
Int[x_*Cos[d_*x^2]*FresnelC[b_*x],x_Symbol] :=
  Sin[d*x^2]*FresnelC[b*x]/(2*d) - b/(4*d)*Int[Sin[2*d*x^2],x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

2: $\int x^m \sin[d x^2] \text{FresnelS}[b x] dx$ when $d^2 = \frac{\pi^2}{4} b^4 \wedge m-1 \in \mathbb{Z}^+$

Derivation: Integration by parts and algebraic simplification

- **Basis:** $-\partial_x \frac{\cos[d x^2]}{2 d} = x \sin[d x^2]$
- **Basis:** If $d^2 = \frac{\pi^2}{4} b^4$, then $\cos[d x^2] \sin[\frac{1}{2} b^2 \pi x^2] = \frac{d}{b^2 \pi} \sin[2 d x^2]$
- **Rule:** If $d^2 = \frac{\pi^2}{4} b^4 \wedge m-1 \in \mathbb{Z}^+$, then

$$\int x^m \sin[d x^2] \text{FresnelS}[b x] dx \rightarrow$$

$$- \frac{x^{m-1} \cos[d x^2] \text{FresnelS}[b x]}{2 d} + \frac{1}{2 b \pi} \int x^{m-1} \sin[2 d x^2] dx + \frac{m-1}{2 d} \int x^{m-2} \cos[d x^2] \text{FresnelS}[b x] dx$$

- **Basis:** $\partial_x \frac{\sin[d x^2]}{2 d} = x \cos[d x^2]$
- **Basis:** If $d^2 = \frac{\pi^2}{4} b^4$, then $\sin[d x^2] \cos[\frac{1}{2} b^2 \pi x^2] = \frac{1}{2} \sin[2 d x^2]$
- **Rule:** If $d^2 = \frac{\pi^2}{4} b^4 \wedge m-1 \in \mathbb{Z}^+$, then

$$\int x^m \cos[d x^2] \text{FresnelC}[b x] dx \rightarrow$$

$$\frac{x^{m-1} \sin[d x^2] \text{FresnelC}[b x]}{2 d} - \frac{b}{4 d} \int x^{m-1} \sin[2 d x^2] dx - \frac{m-1}{2 d} \int x^{m-2} \sin[d x^2] \text{FresnelC}[b x] dx$$

Program code:

```
Int[x_^m * Sin[d.*x_^2] * FresnelS[b.*x_], x_Symbol] :=
  -x^(m-1) * Cos[d*x^2] * FresnelS[b*x] / (2*d) +
  1 / (2*b*Pi) * Int[x^(m-1) * Sin[2*d*x^2], x] +
  (m-1) / (2*d) * Int[x^(m-2) * Cos[d*x^2] * FresnelS[b*x], x] /;
FreeQ[{b,d},x] && EqQ[d^2, Pi^2/4*b^4] && IGtQ[m,1]
```

```
Int[x_^m * Cos[d.*x_^2] * FresnelC[b.*x_], x_Symbol] :=
  x^(m-1) * Sin[d*x^2] * FresnelC[b*x] / (2*d) -
  b / (4*d) * Int[x^(m-1) * Sin[2*d*x^2], x] -
  (m-1) / (2*d) * Int[x^(m-2) * Sin[d*x^2] * FresnelC[b*x], x] /;
FreeQ[{b,d},x] && EqQ[d^2, Pi^2/4*b^4] && IGtQ[m,1]
```

2: $\int x^m \sin[d x^2] \text{FresnelS}[b x] dx$ when $d^2 = \frac{\pi^2}{4} b^4 \bigwedge m+2 \in \mathbb{Z}^-$

Derivation: Inverted integration by parts

■ **Rule:** If $d^2 = \frac{\pi^2}{4} b^4 \bigwedge m+2 \in \mathbb{Z}^-$, then

$$\int x^m \sin[d x^2] \text{FresnelS}[b x] dx \rightarrow \frac{x^{m+1} \sin[d x^2] \text{FresnelS}[b x]}{m+1} - \frac{d x^{m+2}}{\pi b (m+1) (m+2)} + \frac{d}{\pi b (m+1)} \int x^{m+1} \cos[2 d x^2] dx - \frac{2 d}{m+1} \int x^{m+2} \cos[d x^2] \text{FresnelS}[b x] dx$$

$$\int x^m \cos[d x^2] \text{FresnelC}[b x] dx \rightarrow \frac{x^{m+1} \cos[d x^2] \text{FresnelC}[b x]}{m+1} - \frac{b x^{m+2}}{2 (m+1) (m+2)} - \frac{b}{2 (m+1)} \int x^{m+1} \cos[2 d x^2] dx + \frac{2 d}{m+1} \int x^{m+2} \sin[d x^2] \text{FresnelC}[b x] dx$$

Program code:

```
Int[x^m * Sin[d * x^2] * FresnelS[b * x], x_Symbol] :=
  x^(m+1) * Sin[d * x^2] * FresnelS[b * x] / (m+1) -
  d * x^(m+2) / (Pi * b * (m+1) * (m+2)) +
  d / (Pi * b * (m+1)) * Int[x^(m+1) * Cos[2 * d * x^2], x] -
  2 * d / (m+1) * Int[x^(m+2) * Cos[d * x^2] * FresnelS[b * x], x] /;
FreeQ[{b, d}, x] && EqQ[d^2, Pi^2/4 * b^4] && ILtQ[m, -2]
```

```
Int[x^m * Cos[d * x^2] * FresnelC[b * x], x_Symbol] :=
  x^(m+1) * Cos[d * x^2] * FresnelC[b * x] / (m+1) -
  b * x^(m+2) / (2 * (m+1) * (m+2)) -
  b / (2 * (m+1)) * Int[x^(m+1) * Cos[2 * d * x^2], x] +
  2 * d / (m+1) * Int[x^(m+2) * Sin[d * x^2] * FresnelC[b * x], x] /;
FreeQ[{b, d}, x] && EqQ[d^2, Pi^2/4 * b^4] && ILtQ[m, -2]
```

X: $\int (e x)^m \sin[c + d x^2] \text{FresnelS}[a + b x]^n dx$

Rule:

$$\int (e x)^m \sin[c + d x^2] \text{FresnelS}[a + b x]^n dx \rightarrow \int (e x)^m \sin[c + d x^2] \text{FresnelS}[a + b x]^n dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sin[c_.+d_.*x_^2]*FresnelS[a_.+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[(e*x)^m*Sin[c+d*x^2]*FresnelS[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

```
Int[(e_.*x_)^m_.*Cos[c_.+d_.*x_^2]*FresnelC[a_.+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[(e*x)^m*Cos[c+d*x^2]*FresnelC[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

$$7. \int (e x)^m \cos[c + d x^2] \operatorname{FresnelS}[a + b x]^n dx$$

$$1. \int x^m \cos[d x^2] \operatorname{FresnelS}[b x] dx \text{ when } d^2 = \frac{\pi^2}{4} b^4 \bigwedge m \in \mathbb{Z}$$

$$1. \int x^m \cos[d x^2] \operatorname{FresnelS}[b x] dx \text{ when } d^2 = \frac{\pi^2}{4} b^4 \bigwedge m \in \mathbb{Z}^+$$

$$\textcolor{red}{1}: \int x \cos[d x^2] \operatorname{FresnelS}[b x] dx \text{ when } d^2 = \frac{\pi^2}{4} b^4$$

Derivation: Integration by parts and algebraic simplification

- **Basis:** $\partial_x \frac{\sin[d x^2]}{2 d} = x \cos[d x^2]$
- **Basis:** If $d^2 = \frac{\pi^2}{4} b^4$, then $\sin[d x^2] \sin\left[\frac{1}{2} b^2 \pi x^2\right] = \frac{2 d}{\pi b^2} \sin[d x^2]^2$
- **Rule:** If $d^2 = \frac{\pi^2}{4} b^4$, then

$$\int x \cos[d x^2] \operatorname{FresnelS}[b x] dx \rightarrow \frac{\sin[d x^2] \operatorname{FresnelS}[b x]}{2 d} - \frac{1}{\pi b} \int \sin[d x^2]^2 dx$$

- **Basis:** $-\partial_x \frac{\cos[d x^2]}{2 d} = x \sin[d x^2]$
- **Basis:** If $d^2 = \frac{\pi^2}{4} b^4$, then $\cos[d x^2] \cos\left[\frac{1}{2} b^2 \pi x^2\right] = \cos[d x^2]^2$
- **Rule:** If $d^2 = \frac{\pi^2}{4} b^4$, then

$$\int x \sin[d x^2] \operatorname{FresnelC}[b x] dx \rightarrow -\frac{\cos[d x^2] \operatorname{FresnelC}[b x]}{2 d} + \frac{b}{2 d} \int \cos[d x^2]^2 dx$$

Program code:

```
Int[x_*Cos[d_*x^2]*FresnelS[b_*x],x_Symbol] :=
  Sin[d*x^2]*FresnelS[b*x]/(2*d) - 1/(Pi*b)*Int[Sin[d*x^2]^2,x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

```
Int[x_*Sin[d_*x^2]*FresnelC[b_*x],x_Symbol] :=
  -Cos[d*x^2]*FresnelC[b*x]/(2*d) + b/(2*d)*Int[Cos[d*x^2]^2,x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

2: $\int x^m \cos[d x^2] \text{FresnelS}[b x] dx$ when $d^2 = \frac{\pi^2}{4} b^4 \wedge m-1 \in \mathbb{Z}^+$

Derivation: Integration by parts and algebraic simplification

- **Basis:** $\partial_x \frac{\sin[d x^2]}{2 d} = x \cos[d x^2]$
- **Basis:** If $d^2 = \frac{\pi^2}{4} b^4$, then $\sin[d x^2] \sin\left[\frac{1}{2} b^2 \pi x^2\right] = \frac{2 d}{\pi b^2} \sin[d x^2]^2$
- **Rule:** If $d^2 = \frac{\pi^2}{4} b^4 \wedge m-1 \in \mathbb{Z}^+$, then

$$\int x^m \cos[d x^2] \text{FresnelS}[b x] dx \rightarrow \frac{x^{m-1} \sin[d x^2] \text{FresnelS}[b x]}{2 d} - \frac{1}{\pi b} \int x^{m-1} \sin[d x^2]^2 dx - \frac{m-1}{2 d} \int x^{m-2} \sin[d x^2] \text{FresnelS}[b x] dx$$

- **Basis:** $-\partial_x \frac{\cos[d x^2]}{2 d} = x \sin[d x^2]$
- **Basis:** If $d^2 = \frac{\pi^2}{4} b^4$, then $\cos[d x^2] \cos\left[\frac{1}{2} b^2 \pi x^2\right] = \cos[d x^2]^2$
- **Rule:** If $d^2 = \frac{\pi^2}{4} b^4$, then

$$\int x^m \sin[d x^2] \text{FresnelC}[b x] dx \rightarrow -\frac{x^{m-1} \cos[d x^2] \text{FresnelC}[b x]}{2 d} + \frac{b}{2 d} \int x^{m-1} \cos[d x^2]^2 dx + \frac{m-1}{2 d} \int x^{m-2} \cos[d x^2] \text{FresnelC}[b x] dx$$

Program code:

```
Int[x_^m_*Cos[d_*x_^2]*FresnelS[b_*x_],x_Symbol] :=
  x^(m-1)*Sin[d*x^2]*FresnelS[b*x]/(2*d) -
  1/(Pi*b)*Int[x^(m-1)*Sin[d*x^2]^2,x] -
  (m-1)/(2*d)*Int[x^(m-2)*Sin[d*x^2]*FresnelS[b*x],x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4] && IGtQ[m,1]
```

```
Int[x_^m_*Sin[d_*x_^2]*FresnelC[b_*x_],x_Symbol] :=
  -x^(m-1)*Cos[d*x^2]*FresnelC[b*x]/(2*d) +
  b/(2*d)*Int[x^(m-1)*Cos[d*x^2]^2,x] +
  (m-1)/(2*d)*Int[x^(m-2)*Cos[d*x^2]*FresnelC[b*x],x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4] && IGtQ[m,1]
```


2: $\int x^m \cos[d x^2] \text{FresnelS}[b x] dx$ when $d^2 = \frac{\pi^2}{4} b^4 \wedge m+1 \in \mathbb{Z}^-$

Derivation: Inverted integration by parts

■ **Rule:** If $d^2 = \frac{\pi^2}{4} b^4 \wedge m+1 \in \mathbb{Z}^-$, then

$$\int x^m \cos[d x^2] \text{FresnelS}[b x] dx \rightarrow \frac{x^{m+1} \cos[d x^2] \text{FresnelS}[b x]}{m+1} - \frac{d}{\pi b (m+1)} \int x^{m+1} \sin[2 d x^2] dx + \frac{2 d}{m+1} \int x^{m+2} \sin[d x^2] \text{FresnelS}[b x] dx$$

$$\int x^m \sin[d x^2] \text{FresnelC}[b x] dx \rightarrow \frac{x^{m+1} \sin[d x^2] \text{FresnelC}[b x]}{m+1} - \frac{b}{2 (m+1)} \int x^{m+1} \sin[2 d x^2] dx - \frac{2 d}{m+1} \int x^{m+2} \cos[d x^2] \text{FresnelC}[b x] dx$$

Program code:

```
Int[x_^m * Cos[d * x^2] * FresnelS[b * x], x_Symbol] :=
  x^(m+1) * Cos[d * x^2] * FresnelS[b * x] / (m+1) -
  d / (Pi * b * (m+1)) * Int[x^(m+1) * Sin[2 * d * x^2], x] +
  2 * d / (m+1) * Int[x^(m+2) * Sin[d * x^2] * FresnelS[b * x], x] /;
FreeQ[{b, d}, x] && EqQ[d^2, Pi^2/4 * b^4] && ILtQ[m, -1]
```

```
Int[x_^m * Sin[d * x^2] * FresnelC[b * x], x_Symbol] :=
  x^(m+1) * Sin[d * x^2] * FresnelC[b * x] / (m+1) -
  b / (2 * (m+1)) * Int[x^(m+1) * Sin[2 * d * x^2], x] -
  2 * d / (m+1) * Int[x^(m+2) * Cos[d * x^2] * FresnelC[b * x], x] /;
FreeQ[{b, d}, x] && EqQ[d^2, Pi^2/4 * b^4] && ILtQ[m, -1]
```

X: $\int (e x)^m \cos[c + d x^2] \operatorname{FresnelS}[a + b x]^n dx$

Rule:

$$\int (e x)^m \cos[c + d x^2] \operatorname{FresnelS}[a + b x]^n dx \rightarrow \int (e x)^m \cos[c + d x^2] \operatorname{FresnelS}[a + b x]^n dx$$

Program code:

```
Int[(e_.*x_)^m_.*Cos[c_.+d_.*x_^2]*FresnelS[a_.+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[(e*x)^m*cos[c+d*x^2]*FresnelS[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

```
Int[(e_.*x_)^m_.*Sin[c_.+d_.*x_^2]*FresnelC[a_.+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[(e*x)^m*sin[c+d*x^2]*FresnelC[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

8. $\int \operatorname{FresnelS}[d(a + b \log[c x^n])] dx$

1: $\int \operatorname{FresnelS}[d(a + b \log[c x^n])] dx$

Derivation: Integration by parts

■ **Basis:** $\partial_x \operatorname{FresnelS}[d(a + b \log[c x^n])] = \frac{b d n \sin\left[\frac{\pi}{2} (d(a + b \log[c x^n]))^2\right]}{x}$

Rule:

$$\int \operatorname{FresnelS}[d(a + b \log[c x^n])] dx \rightarrow x \operatorname{FresnelS}[d(a + b \log[c x^n])] - b d n \int \sin\left[\frac{\pi}{2} (d(a + b \log[c x^n]))^2\right] dx$$

Program code:

```
Int[FresnelS[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  x*FresnelS[d*(a+b*Log[c*x^n])] - b*d*n*Int[Sin[Pi/2*(d*(a+b*Log[c*x^n]))^2],x] /;
FreeQ[{a,b,c,d,n},x]
```

```
Int[FresnelC[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  x*FresnelC[d*(a+b*Log[c*x^n])] - b*d*n*Int[Cos[Pi/2*(d*(a+b*Log[c*x^n]))^2],x] /;
FreeQ[{a,b,c,d,n},x]
```

2: $\int \frac{\text{FresnelS}[d(a + b \log[c x^n])]}{x} dx$

Derivation: Integration by substitution

Basis: $\frac{F[\log[c x^n]]}{x} = \frac{1}{n} \text{Subst}[F[x], x, \log[c x^n]] \partial_x \log[c x^n]$

Rule:

$$\int \frac{\text{FresnelS}[d(a + b \log[c x^n])]}{x} dx \rightarrow \frac{1}{n} \text{Subst}[\text{FresnelS}[d(a + b x)], x, \log[c x^n]]$$

Program code:

```
Int[F_[d_.*(a_.+b_.*Log[c_.*x_^n_.])]/x_,x_Symbol] :=
  1/n*Subst[F[d*(a+b*x)],x,Log[c*x^n]] /;
FreeQ[{a,b,c,d,n},x] && MemberQ[{FresnelS,FresnelC},F]
```

3: $\int (e x)^m \text{FresnelS}[d(a + b \log[c x^n])] dx$ when $m \neq -1$

Derivation: Integration by parts

Basis: $\partial_x \text{FresnelS}[d(a + b \log[c x^n])] = \frac{b d n \sin\left[\frac{\pi}{2} (d(a + b \log[c x^n]))^2\right]}{x}$

Rule: If $m \neq -1$, then

$$\int (e x)^m \text{FresnelS}[d(a + b \log[c x^n])] dx \rightarrow \frac{(e x)^{m+1} \text{FresnelS}[d(a + b \log[c x^n])]}{e(m+1)} - \frac{b d n}{m+1} \int (e x)^m \sin\left[\frac{\pi}{2} (d(a + b \log[c x^n]))^2\right] dx$$

Program code:

```
Int[(e_.*x_)^m_.*FresnelS[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  (e*x)^(m+1)*FresnelS[d*(a+b*Log[c*x^n])]/(e*(m+1)) -
  b*d*n/(m+1)*Int[(e*x)^m*Sin[Pi/2*(d*(a+b*Log[c*x^n]))^2],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[m,-1]
```

```
Int[(e_.*x_)^m_.*FresnelC[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  (e*x)^(m+1)*FresnelC[d*(a+b*Log[c*x^n])]/(e*(m+1)) -
  b*d*n/(m+1)*Int[(e*x)^m*Cos[Pi/2*(d*(a+b*Log[c*x^n]))^2],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[m,-1]
```