Rules for integrands of the form $(d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p$

0: $\int x^m\,(f+g\,x)^n\,\left(b\,x+c\,x^2\right)\,dx$

- **Rule 1.2.1.4.0: If** $c\,f\,(m+2)-b\,g\,(m+n+3)==0$, **then**

$$\int x^m\,(f+g\,x)^n\,\left(b\,x+c\,x^2\right)\,dx\;\longrightarrow\;\frac{c\,x^{m+2}\,(f+g\,x)^{n+1}}{g\,(m+n+3)}$$

- **Program code:**

```
Int[x_^m_.*(f_+g_.*x_)^n_.*(b_.*x_+c_.*x_^2),x_Symbol] :=
  c*x^(m+2)*(f+g*x)^(n+1)/(g*(m+n+3)) /;
FreeQ[{b,c,f,g,m,n},x] && EqQ[c*f*(m+2)-b*g*(m+n+3),0] && NeQ[m+n+3,0]
```

1: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq0\,\wedge\,b^2-4\,a\,c==0\,\wedge\,p\notin\mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $b^2-4\,a\,c==0$, **then** $\partial_x\frac{\left(a+b\,x+c\,x^2\right)^p}{\left(\frac{b}{2}+c\,x\right)^{2\,p}}==0$
- **Rule 1.2.1.4.1: If** $e\,f-d\,g\neq0\,\wedge\,b^2-4\,a\,c==0\,\wedge\,p\notin\mathbb{Z}$, **then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow\;\frac{\left(a+b\,x+c\,x^2\right)^{\text{FracPart}[p]}}{c^{\text{IntPart}[p]}\,\left(\frac{b}{2}+c\,x\right)^{2\,\text{FracPart}[p]}}\int (d+e\,x)^m\,(f+g\,x)^n\,\left(\frac{b}{2}+c\,x\right)^{2\,p}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (a+b*x+c*x^2)^FracPart[p]/(c^IntPart[p]*(b/2+c*x)^(2*FracPart[p]))*Int[(d+e*x)^m*(f+g*x)^n*(b/2+c*x)^(2*p),x] /;
FreeQ[{a,b,c,d,e,f,g,m,n},x] && NeQ[e*f-d*g,0] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

2. $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2=0$

1: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2=0\,\wedge\,p\in\mathbb{Z}$

- **Derivation: Algebraic simplification**
- **Basis: If** $c\,d^2-b\,d\,e+a\,e^2=0$**, then** $a+b\,x+c\,x^2=(d+e\,x)\left(\frac{a}{d}+\frac{c\,x}{e}\right)$
- **Rule 1.2.1.4.2.1: If** $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2=0\,\wedge\,p\in\mathbb{Z}$**, then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\int (d+e\,x)^{m+p}\,(f+g\,x)^n\,\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[(d+e*x)^(m+p)*(f+g*x)^n*(a/d+c/e*x)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[p] && Not[IGtQ[n,0]]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[(d+e*x)^(m+p)*(f+g*x)^n*(a/d+c/e*x)^p,x] /;
FreeQ[{a,c,d,e,f,g,m,n},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] && (IntegerQ[p] || GtQ[a,0] && GtQ[d,0] && EqQ[m+p,0])
```

2. $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}$

1: $\int \frac{x^n\,\left(a+b\,x+c\,x^2\right)^p}{d+e\,x}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $c\,d^2-b\,d\,e+a\,e^2==0$**, then** $\frac{a+b\,x+c\,x^2}{d+e\,x}==\frac{a}{d}+\frac{c\,x}{e}$

Rule 1.2.1.4.2.2.1: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,p>0$**, then**

$$\int \frac{x^n\,\left(a+b\,x+c\,x^2\right)^p}{d+e\,x}\,dx\,\longrightarrow\,\int x^n\,\left(\frac{a}{d}+\frac{c\,x}{e}\right)\,\left(a+b\,x+c\,x^2\right)^{p-1}\,dx$$

Program code:

```
Int[x_^n_.*(a_.+b_.*x_+c_.*x_^2)^p_/(d_+e_.*x_),x_Symbol] :=
  Int[x^n*(a/d+c*x/e)*(a+b*x+c*x^2)^(p-1),x] /;
FreeQ[{a,b,c,d,e,n,p},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] &&
  (Not[IntegerQ[n]] || Not[IntegerQ[2*p]] || IGtQ[n,2] || GtQ[p,0] && NeQ[n,2])
```

```
Int[x_^n_.*(a_+c_.*x_^2)^p_/(d_+e_.*x_),x_Symbol] :=
  Int[x^n*(a/d+c*x/e)*(a+c*x^2)^(p-1),x] /;
FreeQ[{a,c,d,e,n,p},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] &&
  (Not[IntegerQ[n]] || Not[IntegerQ[2*p]] || IGtQ[n,2] || GtQ[p,0] && NeQ[n,2])
```

2: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m\in\mathbb{Z}^-$

Derivation: Algebraic simplification

- **Basis: If** $c\,d^2-b\,d\,e+a\,e^2==0$, **then** $d+e\,x==\frac{a+b\,x+c\,x^2}{\frac{a}{d}+\frac{c\,x}{e}}$
- **Basis: If** $c\,d^2+a\,e^2==0$, **then** $d+e\,x==\frac{d^2\,(a+c\,x^2)}{a\,(d-e\,x)}$

Note: Since $\left(\frac{a}{d}+\frac{c\,x}{e}\right)^{-m}$ **is a polynomial, this rule transforms integrand into an expression of the form** $(d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p$ **for which there are rules.**

Rule 1.2.1.4.2.2.2: If $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m\in\mathbb{Z}^-$, **then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\int\left(\frac{a}{d}+\frac{c\,x}{e}\right)^{-m}\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^{m+p}\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[(a/d+c*x/e)^(-m)*(f+g*x)^n*(a+b*x+c*x^2)^(m+p),x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] && ILtQ[m,0] &&
  (LtQ[n,0] || GtQ[p,0])
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  d^(2*m)/a^m*Int[(f+g*x)^n*(a+c*x^2)^(m+p)/(d-e*x)^m,x] /;
FreeQ[{a,c,d,e,f,g,n,p},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && EqQ[f,0] && ILtQ[m,-1] &&
  Not[IGtQ[n,0] && ILtQ[m+n,0] && Not[GtQ[p,1]]]
```

```
Int[(d_+e_.*x_)^m_*(f_+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  d^(2*m)/a^m*Int[(f+g*x)^n*(a+c*x^2)^(m+p)/(d-e*x)^m,x] /;
FreeQ[{a,c,d,e,f,g,n,p},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && ILtQ[m,0] && IntegerQ[n]
```

3. $\int \frac{(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p}{d+e\,x}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,n\in\mathbb{Z}\,\wedge\,n+2\,p\in\mathbb{Z}^-$

1: $\int \frac{(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p}{d+e\,x}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,n\in\mathbb{Z}^+\wedge\,n+2\,p\in\mathbb{Z}^-$

Derivation: Algebraic simplification and quadratic recurrence 2a

- **Basis: If** $c\,d^2-b\,d\,e+a\,e^2==0$**, then** $\frac{a+b\,x+c\,x^2}{d+e\,x}==\frac{a\,e+c\,d\,x}{d\,e}$

- **Rule 1.2.1.4.2.2.3.1: If** $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,n\in\mathbb{Z}^+\wedge\,n+2\,p\in\mathbb{Z}^-$**, then**

$$\int \frac{(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p}{d+e\,x}\,dx\,\longrightarrow\,\frac{1}{d\,e}\int(a\,e+c\,d\,x)\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^{p-1}\,dx\,\longrightarrow$$

$$-\frac{(2\,c\,d-b\,e)\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^{p+1}}{e\,p\,\left(b^2-4\,a\,c\right)\,(d+e\,x)}-$$

$$\frac{1}{d\,e\,p\,\left(b^2-4\,a\,c\right)}\int(f+g\,x)^{n-1}\,\left(a+b\,x+c\,x^2\right)^p\,(b\,(a\,e\,g\,n-c\,d\,f\,(2\,p+1))-2\,a\,c\,(d\,g\,n-e\,f\,(2\,p+1))-c\,g\,(b\,d-2\,a\,e)\,(n+2\,p+1)\,x)\,dx$$

- **Program code:**

```
Int[(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_/(d_+e_.*x_),x_Symbol] :=
  -(2*c*d-b*e)*(f+g*x)^n*(a+b*x+c*x^2)^(p+1)/(e*p*(b^2-4*a*c)*(d+e*x)) -
  1/(d*e*p*(b^2-4*a*c))*Int[(f+g*x)^(n-1)*(a+b*x+c*x^2)^p*
    Simp[b*(a*e*g*n-c*d*f*(2*p+1))-2*a*c*(d*g*n-e*f*(2*p+1))-c*g*(b*d-2*a*e)*(n+2*p+1)*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] && IGtQ[n,0] && ILtQ
```

```
Int[(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_/(d_+e_.*x_),x_Symbol] :=
  d*(f+g*x)^n*(a+c*x^2)^(p+1)/(2*a*e*p*(d+e*x)) -
  1/(2*d*e*p)*Int[(f+g*x)^(n-1)*(a+c*x^2)^p*Simp[d*g*n-e*f*(2*p+1)-e*g*(n+2*p+1)*x,x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && IGtQ[n,0] && ILtQ[n+2*p,0]
```

2: $\int \frac{(f+g x)^n \left(a+b x+c x^2\right)^p}{d+e x} \, dx$ when $e f-d g \neq 0 \land b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \notin \mathbb{Z} \land n \in \mathbb{Z}^- \land n+2 p \in \mathbb{Z}^-$

Derivation: Algebraic simplification and quadratic recurrence 2b

- **Basis: If** $c d^2-b d e+a e^2 == 0$, **then** $\frac{a+b x+c x^2}{d+e x} == \frac{a e+c d x}{d e}$

Rule 1.2.1.4.2.2.3.2: If $e f-d g \neq 0 \land b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \notin \mathbb{Z} \land n \in \mathbb{Z}^- \land n+2 p \in \mathbb{Z}^-$, **then**

$$\int \frac{(f+g x)^n \left(a+b x+c x^2\right)^p}{d+e x} \, dx \longrightarrow \frac{1}{d e} \int (a e+c d x) (f+g x)^n \left(a+b x+c x^2\right)^{p-1} dx \longrightarrow$$

$$\frac{(f+g x)^{n+1} \left(a+b x+c x^2\right)^p (c d-b e-c e x)}{p (2 c d-b e) (e f-d g)} +$$

$$\frac{1}{p (2 c d-b e) (e f-d g)} \int (f+g x)^n \left(a+b x+c x^2\right)^p (b e g (n+p+1)+c e f (2 p+1)-c d g (n+2 p+1)+c e g (n+2 p+2) x) \, dx$$

Program code:

```
Int[(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_/(d_+e_.*x_),x_Symbol] :=
  (f+g*x)^(n+1)*(a+b*x+c*x^2)^p*(c*d-b*e-c*e*x)/(p*(2*c*d-b*e)*(e*f-d*g)) +
  1/(p*(2*c*d-b*e)*(e*f-d*g))*Int[(f+g*x)^n*(a+b*x+c*x^2)^p*(b*e*g*(n+p+1)+c*e*f*(2*p+1)-c*d*g*(n+2*p+1)+c*e*g*(n+2*p+2)*x),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] &&
  ILtQ[n,0] && ILtQ[n+2*p,0] && Not[IGtQ[n,0]]
```

```
Int[(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_/(d_+e_.*x_),x_Symbol] :=
  d*(f+g*x)^(n+1)*(a+c*x^2)^(p+1)/(2*a*p*(e*f-d*g)*(d+e*x)) +
  1/(p*(2*c*d)*(e*f-d*g))*Int[(f+g*x)^n*(a+c*x^2)^p*(c*e*f*(2*p+1)-c*d*g*(n+2*p+1)+c*e*g*(n+2*p+2)*x),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] &&
  ILtQ[n,0] && ILtQ[n+2*p,0] && Not[IGtQ[n,0]]
```

4. $\int (d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p dx$ when $e f-d g \neq 0 \land b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \notin \mathbb{Z} \land m+p == 0$

1: $\int (d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p dx$ when
$e f-d g \neq 0 \land b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \notin \mathbb{Z} \land m+p == 0 \land c e f+c d g-b e g == 0 \land m-n-1 \neq 0$

Rule 1.2.1.4.2.2.4.1: If $e f-d g \neq 0 \land b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \notin \mathbb{Z} \land m+p == 0 \land c e f+c d g-b e g == 0 \land m-n-1 \neq 0$, **then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx \,\longrightarrow\, -\frac{e\,(d+e\,x)^{m-1}\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^{p+1}}{c\,(m-n-1)}$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  -e*(d+e*x)^(m-1)*(f+g*x)^n*(a+b*x+c*x^2)^(p+1)/(c*(m-n-1)) /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p,0] && EqQ[c*e*f+c*d*g-b*e*g,0] && NeQ[m-n-1,0]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  -e*(d+e*x)^(m-1)*(f+g*x)^n*(a+c*x^2)^(p+1)/(c*(m-n-1)) /;
FreeQ[{a,c,d,e,f,g,m,n,p},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p,0] && EqQ[e*f+d*g,0] && NeQ[m-n-1,0]
```

2: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m+p==0\,\wedge\,m-n-2==0$

Rule 1.2.1.4.2.2.4.2: If $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m+p==0\,\wedge\,m-n-2==0$**, then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx \,\longrightarrow\, -\frac{e^2\,(d+e\,x)^{m-1}\,(f+g\,x)^{n+1}\,\left(a+b\,x+c\,x^2\right)^{p+1}}{(n+1)\,(c\,e\,f+c\,d\,g-b\,e\,g)}$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  -e^2*(d+e*x)^(m-1)*(f+g*x)^(n+1)*(a+b*x+c*x^2)^(p+1)/((n+1)*(c*e*f+c*d*g-b*e*g)) /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] && EqQ[m+p,0]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  -e^2*(d+e*x)^(m-1)*(f+g*x)^(n+1)*(a+c*x^2)^(p+1)/(c*(n+1)*(e*f+d*g)) /;
FreeQ[{a,c,d,e,f,g,m,n,p},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && EqQ[m+p,0] && EqQ[m-n-2,0]
```

3. $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m+p==0\,\wedge\,p>0$

1:

$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m+p==0\,\wedge\,p>0\,\wedge\,n<-1$

Rule 1.2.1.4.2.2.4.3.1: If $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m+p==0\,\wedge\,p>0\,\wedge\,n<-1$**, then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx \longrightarrow$$

$$\frac{(d+e\,x)^m\,(f+g\,x)^{n+1}\,\left(a+b\,x+c\,x^2\right)^p}{g\,(n+1)}+\frac{c\,m}{e\,g\,(n+1)}\int (d+e\,x)^{m+1}\,(f+g\,x)^{n+1}\,\left(a+b\,x+c\,x^2\right)^{p-1}\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^m*(f+g*x)^(n+1)*(a+b*x+c*x^2)^p/(g*(n+1)) +
  c*m/(e*g*(n+1))*Int[(d+e*x)^(m+1)*(f+g*x)^(n+1)*(a+b*x+c*x^2)^(p-1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p,0] && GtQ[p,0] && LtQ[n,-1] && Not[IntegerQ[n+p] && LeQ[n+p+2,0]]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^m*(f+g*x)^(n+1)*(a+c*x^2)^p/(g*(n+1)) +
  c*m/(e*g*(n+1))*Int[(d+e*x)^(m+1)*(f+g*x)^(n+1)*(a+c*x^2)^(p-1),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p,0] && GtQ[p,0] && LtQ[n,-1] && Not[IntegerQ[n+p] && LeQ[n+p+2,0]]
```

2: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\land\,b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,m+p==0\,\land\,p>0\,\land\,m-n-1\neq 0$

- Rule 1.2.1.4.2.2.4.3.2: If $e\,f-d\,g\neq 0\,\land\,b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,m+p==0\,\land\,p>0\,\land\,m-n-1\neq 0$, then

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx \longrightarrow$$

$$-\frac{(d+e\,x)^m\,(f+g\,x)^{n+1}\,\left(a+b\,x+c\,x^2\right)^p}{g\,(m-n-1)}-\frac{m\,(c\,e\,f+c\,d\,g-b\,e\,g)}{e^2\,g\,(m-n-1)}\int (d+e\,x)^{m+1}\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^{p-1}\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  -(d+e*x)^m*(f+g*x)^(n+1)*(a+b*x+c*x^2)^p/(g*(m-n-1)) -
  m*(c*e*f+c*d*g-b*e*g)/(e^2*g*(m-n-1))*Int[(d+e*x)^(m+1)*(f+g*x)^n*(a+b*x+c*x^2)^(p-1),x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p,0] && GtQ[p,0] && NeQ[m-n-1,0] && Not[IGtQ[n,0]] && Not[IntegerQ[n+p] && LtQ[n+p+2,0]] && RationalQ[n
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  -(d+e*x)^m*(f+g*x)^(n+1)*(a+c*x^2)^p/(g*(m-n-1)) -
  c*m*(e*f+d*g)/(e^2*g*(m-n-1))*Int[(d+e*x)^(m+1)*(f+g*x)^n*(a+c*x^2)^(p-1),x] /;
FreeQ[{a,c,d,e,f,g,n},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p,0] && GtQ[p,0] && NeQ[m-n-1,0] && Not[IGtQ[n,0]] && Not[IntegerQ[n+p] && LtQ[n+p+2,0]] && RationalQ[n
```

4. $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\land\,b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2=0\,\land\,p\notin\mathbb{Z}\,\land\,m+p=0\,\land\,p<-1$

1: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\land\,b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2=0\,\land\,p\notin\mathbb{Z}\,\land\,m+p=0\,\land\,p<-1\,\land\,n>0$

Rule 1.2.1.4.2.2.4.4.1: If $e\,f-d\,g\neq 0\,\land\,b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2=0\,\land\,p\notin\mathbb{Z}\,\land\,m+p=0\,\land\,p<-1\,\land\,n>0$, **then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow$$

$$\frac{e\,(d+e\,x)^{m-1}\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^{p+1}}{c\,(p+1)}-\frac{e\,g\,n}{c\,(p+1)}\int (d+e\,x)^{m-1}\,(f+g\,x)^{n-1}\,\left(a+b\,x+c\,x^2\right)^{p+1}\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m-1)*(f+g*x)^n*(a+b*x+c*x^2)^(p+1)/(c*(p+1)) -
  e*g*n/(c*(p+1))*Int[(d+e*x)^(m-1)*(f+g*x)^(n-1)*(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p,0] && LtQ[p,-1] && GtQ[n,0]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  e*(d+e*x)^(m-1)*(f+g*x)^n*(a+c*x^2)^(p+1)/(c*(p+1)) -
  e*g*n/(c*(p+1))*Int[(d+e*x)^(m-1)*(f+g*x)^(n-1)*(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p,0] && LtQ[p,-1] && GtQ[n,0]
```

2: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\land\,b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2=0\,\land\,p\notin\mathbb{Z}\,\land\,m+p=0\,\land\,p<-1$

Rule 1.2.1.4.2.2.4.4.2: If $e\,f-d\,g\neq 0\,\land\,b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2=0\,\land\,p\notin\mathbb{Z}\,\land\,m+p=0\,\land\,p<-1$, **then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow$$

$$\frac{e^2 (d+e x)^{m-1} (f+g x)^{n+1} \left(a+b x+c x^2\right)^{p+1}}{(p+1) (c e f+c d g-b e g)}+\frac{e^2 g (m-n-2)}{(p+1) (c e f+c d g-b e g)}\int (d+e x)^{m-1} (f+g x)^n \left(a+b x+c x^2\right)^{p+1} dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e^2*(d+e*x)^(m-1)*(f+g*x)^(n+1)*(a+b*x+c*x^2)^(p+1)/((p+1)*(c*e*f+c*d*g-b*e*g)) +
  e^2*g*(m-n-2)/((p+1)*(c*e*f+c*d*g-b*e*g))*Int[(d+e*x)^(m-1)*(f+g*x)^n*(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] && EqQ[m+p,0] &&
  LtQ[p,-1] && RationalQ[n]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  e^2*(d+e*x)^(m-1)*(f+g*x)^(n+1)*(a+c*x^2)^(p+1)/(c*(p+1)*(e*f+d*g)) +
  e^2*g*(m-n-2)/(c*(p+1)*(e*f+d*g))*Int[(d+e*x)^(m-1)*(f+g*x)^n*(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e,f,g,n},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && EqQ[m+p,0] && LtQ[p,-1] && RationalQ[n]
```

5:

$\int (d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p dx$ when $e f-d g\neq 0 \land b^2-4 a c\neq 0 \land c d^2-b d e+a e^2==0 \land p\notin \mathbb{Z} \land m+p==0 \land n>0 \land m-n-1\neq 0$

–

Rule 1.2.1.4.2.2.4.5: If $e f-d g\neq 0 \land b^2-4 a c\neq 0 \land c d^2-b d e+a e^2==0 \land p\notin \mathbb{Z} \land m+p==0 \land n>0 \land m-n-1\neq 0$**, then**

$$\int (d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p dx \longrightarrow$$

$$-\frac{e (d+e x)^{m-1} (f+g x)^n \left(a+b x+c x^2\right)^{p+1}}{c (m-n-1)}-\frac{n (c e f+c d g-b e g)}{c e (m-n-1)}\int (d+e x)^m (f+g x)^{n-1} \left(a+b x+c x^2\right)^p dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  -e*(d+e*x)^(m-1)*(f+g*x)^n*(a+b*x+c*x^2)^(p+1)/(c*(m-n-1)) -
  n*(c*e*f+c*d*g-b*e*g)/(c*e*(m-n-1))*Int[(d+e*x)^m*(f+g*x)^(n-1)*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p,0] && GtQ[n,0] && NeQ[m-n-1,0] && (IntegerQ[2*p] || IntegerQ[n])
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  -e*(d+e*x)^(m-1)*(f+g*x)^n*(a+c*x^2)^(p+1)/(c*(m-n-1)) -
  n*(e*f+d*g)/(e*(m-n-1))*Int[(d+e*x)^m*(f+g*x)^(n-1)*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,f,g,m,p},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p,0] && GtQ[n,0] && NeQ[m-n-1,0] && (IntegerQ[2*p] || IntegerQ[n])
```

6: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m+p==0\,\wedge\,n<-1$

Rule 1.2.1.4.2.2.4.6: If $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m+p==0\,\wedge\,n<-1$**, then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$

$$-\frac{e^2\,(d+e\,x)^{m-1}\,(f+g\,x)^{n+1}\,\left(a+b\,x+c\,x^2\right)^{p+1}}{(n+1)\,(c\,e\,f+c\,d\,g-b\,e\,g)}-\frac{c\,e\,(m-n-2)}{(n+1)\,(c\,e\,f+c\,d\,g-b\,e\,g)}\int (d+e\,x)^m\,(f+g\,x)^{n+1}\,\left(a+b\,x+c\,x^2\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  -e^2*(d+e*x)^(m-1)*(f+g*x)^(n+1)*(a+b*x+c*x^2)^(p+1)/((n+1)*(c*e*f+c*d*g-b*e*g)) -
  c*e*(m-n-2)/((n+1)*(c*e*f+c*d*g-b*e*g))*Int[(d+e*x)^m*(f+g*x)^(n+1)*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p,0] && LtQ[n,-1] && IntegerQ[2*p]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  -e^2*(d+e*x)^(m-1)*(f+g*x)^(n+1)*(a+c*x^2)^(p+1)/((n+1)*(c*e*f+c*d*g)) -
  e*(m-n-2)/((n+1)*(e*f+d*g))*Int[(d+e*x)^m*(f+g*x)^(n+1)*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,f,g,m,p},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p,0] && LtQ[n,-1] && IntegerQ[2*p]
```

7: $\int \frac{\sqrt{d+e\,x}}{(f+g\,x)\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0$

Derivation: Integration by substitution

- Basis: If $c\,d^2-b\,d\,e+a\,e^2==0$, then $\frac{\sqrt{d+e\,x}}{x\,\sqrt{a+b\,x+c\,x^2}}==-2\,d\,\text{Subst}\left[\frac{1}{a-d\,x^2},\,x,\,\frac{\sqrt{a+b\,x+c\,x^2}}{\sqrt{d+e\,x}}\right]\partial_x\frac{\sqrt{a+b\,x+c\,x^2}}{\sqrt{d+e\,x}}$
- Basis: If $c\,d^2-b\,d\,e+a\,e^2==0$, then $\frac{\sqrt{d+e\,x}}{(f+g\,x)\,\sqrt{a+b\,x+c\,x^2}}==2\,e^2\,\text{Subst}\left[\frac{1}{c\,(e\,f+d\,g)-b\,e\,g+e^2\,g\,x^2},\,x,\,\frac{\sqrt{a+b\,x+c\,x^2}}{\sqrt{d+e\,x}}\right]\partial_x\frac{\sqrt{a+b\,x+c\,x^2}}{\sqrt{d+e\,x}}$
- Rule 1.2.1.4.2.2.4.7: If $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0$, then

$$\int \frac{\sqrt{d+e\,x}}{(f+g\,x)\,\sqrt{a+b\,x+c\,x^2}}\,dx\,\rightarrow\,2\,e^2\,\text{Subst}\left[\int\frac{1}{c\,(e\,f+d\,g)-b\,e\,g+e^2\,g\,x^2}\,dx,\,x,\,\frac{\sqrt{a+b\,x+c\,x^2}}{\sqrt{d+e\,x}}\right]$$

Program code:

```
Int[Sqrt[d_+e_.*x_]/((f_.+g_.*x_)*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  2*e^2*Subst[Int[1/(c*(e*f+d*g)-b*e*g+e^2*g*x^2),x],x,Sqrt[a+b*x+c*x^2]/Sqrt[d+e*x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[Sqrt[d_+e_.*x_]/((f_.+g_.*x_)*Sqrt[a_+c_.*x_^2]),x_Symbol] :=
  2*e^2*Subst[Int[1/(c*(e*f+d*g)+e^2*g*x^2),x],x,Sqrt[a+c*x^2]/Sqrt[d+e*x]] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0]
```

5. $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m+p-1==0$

1: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when

$e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m+p-1==0\,\wedge\,b\,e\,g\,(n+1)+c\,e\,f\,(p+1)-c\,d\,g\,(2\,n+p+3)==0\,\wedge\,n+p+2\neq 0$

Rule 1.2.1.4.2.2.5.1: If $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge$ $m+p-1==0\,\wedge\,b\,e\,g\,(n+1)+c\,e\,f\,(p+1)-c\,d\,g\,(2\,n+p+3)==0\,\wedge\,n+p+2\neq 0$**, then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\frac{e^2\,(d+e\,x)^{m-2}\,(f+g\,x)^{n+1}\,\left(a+b\,x+c\,x^2\right)^{p+1}}{c\,g\,(n+p+2)}$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e^2*(d+e*x)^(m-2)*(f+g*x)^(n+1)*(a+b*x+c*x^2)^(p+1)/(c*g*(n+p+2)) /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p-1,0] && EqQ[b*e*g*(n+1)+c*e*f*(p+1)-c*d*g*(2*n+p+3),0] && NeQ[n+p+2,0]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  e^2*(d+e*x)^(m-2)*(f+g*x)^(n+1)*(a+c*x^2)^(p+1)/(c*g*(n+p+2)) /;
FreeQ[{a,c,d,e,f,g,m,n,p},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p-1,0] && EqQ[e*f*(p+1)-d*g*(2*n+p+3),0] && NeQ[n+p+2,0]
```

2: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\land\,b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,m+p-1==0\,\land\,n<-1$

Rule 1.2.1.4.2.2.5.2: If $e\,f-d\,g\neq 0\,\land\,b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}\,\land\,m+p-1==0\,\land\,n<-1$, then

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$

$$\frac{e^2\,(e\,f-d\,g)\,(d+e\,x)^{m-2}\,(f+g\,x)^{n+1}\,\left(a+b\,x+c\,x^2\right)^{p+1}}{g\,(n+1)\,(c\,e\,f+c\,d\,g-b\,e\,g)}-\frac{e\,(b\,e\,g\,(n+1)+c\,e\,f\,(p+1)-c\,d\,g\,(2\,n+p+3))}{g\,(n+1)\,(c\,e\,f+c\,d\,g-b\,e\,g)}\int (d+e\,x)^{m-1}\,(f+g\,x)^{n+1}\,\left(a+b\,x+c\,x^2\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e^2*(e*f-d*g)*(d+e*x)^(m-2)*(f+g*x)^(n+1)*(a+b*x+c*x^2)^(p+1)/(g*(n+1)*(c*e*f+c*d*g-b*e*g)) -
  e*(b*e*g*(n+1)+c*e*f*(p+1)-c*d*g*(2*n+p+3))/(g*(n+1)*(c*e*f+c*d*g-b*e*g))*
    Int[(d+e*x)^(m-1)*(f+g*x)^(n+1)*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p-1,0] && LtQ[n,-1] && IntegerQ[2*p]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  e^2*(e*f-d*g)*(d+e*x)^(m-2)*(f+g*x)^(n+1)*(a+c*x^2)^(p+1)/(c*g*(n+1)*(e*f+d*g)) -
  e*(e*f*(p+1)-d*g*(2*n+p+3))/(g*(n+1)*(e*f+d*g))*Int[(d+e*x)^(m-1)*(f+g*x)^(n+1)*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,f,g,m,p},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p-1,0] && LtQ[n,-1] && IntegerQ[2*p]
```

3: $\int (d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p dx$ when $e f-d g\neq 0 \land b^2-4 a c\neq 0 \land c d^2-b d e+a e^2==0 \land p\notin \mathbb{Z} \land m+p-1==0 \land n\nless -1$

Rule 1.2.1.4.2.2.5.3: If $e f-d g\neq 0 \land b^2-4 a c\neq 0 \land c d^2-b d e+a e^2==0 \land p\notin \mathbb{Z} \land m+p-1==0 \land n<-1$**, then**

$$\int (d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p dx \longrightarrow$$
$$\frac{e^2 (d+e x)^{m-2} (f+g x)^{n+1} \left(a+b x+c x^2\right)^{p+1}}{c g (n+p+2)}-\frac{b e g (n+1)+c e f (p+1)-c d g (2 n+p+3)}{c g (n+p+2)}\int (d+e x)^{m-1} (f+g x)^n \left(a+b x+c x^2\right)^p dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  e^2*(d+e*x)^(m-2)*(f+g*x)^(n+1)*(a+b*x+c*x^2)^(p+1)/(c*g*(n+p+2)) -
  (b*e*g*(n+1)+c*e*f*(p+1)-c*d*g*(2*n+p+3))/(c*g*(n+p+2))*Int[(d+e*x)^(m-1)*(f+g*x)^n*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p-1,0] && Not[LtQ[n,-1]] && IntegerQ[2*p]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  e^2*(d+e*x)^(m-2)*(f+g*x)^(n+1)*(a+c*x^2)^(p+1)/(c*g*(n+p+2)) -
  (e*f*(p+1)-d*g*(2*n+p+3))/(g*(n+p+2))*Int[(d+e*x)^(m-1)*(f+g*x)^n*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,f,g,m,n,p},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] &&
  Not[IntegerQ[p]] && EqQ[m+p-1,0] && Not[LtQ[n,-1]] && IntegerQ[2*p]
```

6: $\int (d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p dx$ when $e f-d g\neq 0 \land b^2-4 a c\neq 0 \land c d^2-b d e+a e^2==0 \land p\notin \mathbb{Z} \land (m\in \mathbb{Z}^+ \lor (m \mid n)\in \mathbb{Z})$

Derivation: Algebraic expansion

Rule 1.2.1.4.2.2.6: If $e f-d g\neq 0 \land b^2-4 a c\neq 0 \land c d^2-b d e+a e^2==0 \land p\notin \mathbb{Z} \land (m\in \mathbb{Z}^+ \lor (m \mid n)\in \mathbb{Z})$**, then**

$$\int (d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p dx \longrightarrow \int \text{ExpandIntegrand}\left[(d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p, x\right] dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)^n*(a+b*x+c*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IntegerQ[p]] && ILtQ[m,0] && (ILtQ[n,0] || IGtQ[n,0] && ILtQ[p+1/2,0]) && Not[IGtQ[n,0]]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[1/Sqrt[a+c*x^2],(d+e*x)^m*(f+g*x)^n*(a+c*x^2)^(p+1/2),x],x] /;
FreeQ[{a,c,d,e,f,g,n,p},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] && IntegerQ[p-1/2] && ILtQ[m,0] && ILtQ[n,0] && Not[IGtQ[n,0]]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)^n*(a+c*x^2)^p,x],x] /;
FreeQ[{a,c,d,e,f,g,n,p},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && ILtQ[m,0] && (ILtQ[n,0] || IGtQ[n,0] && I
```

7: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\bigwedge\,b^2-4\,a\,c\neq 0\,\bigwedge\,c\,d^2-b\,d\,e+a\,e^2=0\,\bigwedge\,p+\frac{1}{2}\in\mathbb{Z}^-\,\bigwedge\,m\in\mathbb{Z}^+\,\bigwedge\,n\in\mathbb{Z}^+$

Derivation: Algebraic expansion and special quadratic recurrence 2b

Basis: If $c\,d^2-b\,d\,e+a\,e^2=0$**, then** $(d+e\,x)\,(a\,e+c\,d\,x)=d\,e\,\left(a+b\,x+c\,x^2\right)$

Rule 1.2.1.4.2.2.7: If $e\,f-d\,g\neq 0\,\bigwedge\,b^2-4\,a\,c\neq 0\,\bigwedge\,c\,d^2-b\,d\,e+a\,e^2=0\,\bigwedge\,p+\frac{1}{2}\in\mathbb{Z}^-\,\bigwedge\,m\in\mathbb{Z}^+\,\bigwedge\,n\in\mathbb{Z}^+$,
let $Q_{n-1}[x]\to\text{PolynomialQuotient}[(f+g\,x)^n,\,a\,e+c\,d\,x,\,x]$ **and** $h\to\text{PolynomialRemainder}[(f+g\,x)^n,\,a\,e+c\,d\,x,\,x]$**, then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$

$$h\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx+d\,e\int (d+e\,x)^{m-1}\,Q_{n-1}[x]\,\left(a+b\,x+c\,x^2\right)^{p+1}\,dx\,\longrightarrow$$

$$\frac{h\,(2\,c\,d-b\,e)\,(d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^{p+1}}{e\,(p+1)\,\left(b^2-4\,a\,c\right)}+$$
$$\frac{1}{(p+1)\,\left(b^2-4\,a\,c\right)}\int (d+e\,x)^{m-1}\,\left(a+b\,x+c\,x^2\right)^{p+1}\,\left(d\,e\,(p+1)\,\left(b^2-4\,a\,c\right)\,Q_{n-1}[x]-h\,(2\,c\,d-b\,e)\,(m+2\,p+2)\right)\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[(f+g*x)^n,a*e+c*d*x,x], h=PolynomialRemainder[(f+g*x)^n,a*e+c*d*x,x]},
  h*(2*c*d-b*e)*(d+e*x)^m*(a+b*x+c*x^2)^(p+1)/(e*(p+1)*(b^2-4*a*c)) +
  1/((p+1)*(b^2-4*a*c))*Int[(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1)*
    ExpandToSum[d*e*(p+1)*(b^2-4*a*c)*Q-h*(2*c*d-b*e)*(m+2*p+2),x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && ILtQ[p+1/2,0] && IGtQ[m,0] && IGtQ[n,
```

```
Int[(d_.+e_.*x_)^m_.*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[(f+g*x)^n,a*e+c*d*x,x], h=PolynomialRemainder[(f+g*x)^n,a*e+c*d*x,x]},
  -d*h*(d+e*x)^m*(a+c*x^2)^(p+1)/(2*a*e*(p+1)) +
  d/(2*a*(p+1))*Int[(d+e*x)^(m-1)*(a+c*x^2)^(p+1)*ExpandToSum[2*a*e*(p+1)*Q+h*(m+2*p+2),x],x]] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] && ILtQ[p+1/2,0] && IGtQ[m,0] && IGtQ[n,0] && Not[IGtQ[n,0]]
```

8: $\int (d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p dx$ when $e f-d g\neq 0 \land b^2-4 a c\neq 0 \land c d^2-b d e+a e^2==0 \land p\notin\mathbb{Z} \land m+n+2 p+1==0 \land m\in\mathbb{Z}^- \land n\in\mathbb{Z}^-$

Derivation: Algebraic expansion

Rule 1.2.1.4.2.2.8: If $e f-d g\neq 0 \land b^2-4 a c\neq 0 \land c d^2-b d e+a e^2==0 \land p\notin\mathbb{Z} \land m+n+2 p+1==0 \land n\in\mathbb{Z} \land m\in\mathbb{Z}^-$**, then**

$$\int (d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p dx \longrightarrow \int \left(a+b x+c x^2\right)^p \text{ExpandIntegrand}\left[(d+e x)^m (f+g x)^n,\ x\right] dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x+c*x^2)^p,(d+e*x)^m*(f+g*x)^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] &&
  EqQ[m+n+2*p+1,0] && ILtQ[m,0] && ILtQ[n,0]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+c*x^2)^p,(d+e*x)^m*(f+g*x)^n,x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && EqQ[m+n+2*p+1,0] && ILtQ[m,0] && ILtQ[n,0]
```

x: $\int (d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p dx$ when $e f-d g\neq 0 \land b^2-4 a c\neq 0 \land c d^2-b d e+a e^2==0 \land p\notin\mathbb{Z} \land m+n+2 p+1\neq 0 \land n\in\mathbb{Z}^+$

Derivation: Algebraic expansion and quadratic recurrence 3a with $A=d$**,** $B=e$ **and** $m=m-1$

Rule 1.2.1.4.2.2.x: If $e f-d g\neq 0 \land b^2-4 a c\neq 0 \land c d^2-b d e+a e^2==0 \land p\notin\mathbb{Z} \land m+n+2 p+1\neq 0 \land n\in\mathbb{Z}^+$**, then**

$$\int (d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p dx \longrightarrow$$

$$\int (d+e x)^m \left((f+g x)^n-\frac{g^n}{e^n}(d+e x)^n\right)\left(a+b x+c x^2\right)^p dx+\frac{g^n}{e^n}\int (d+e x)^{m+n}\left(a+b x+c x^2\right)^p dx \longrightarrow$$

$$\frac{g^n (d+e x)^{m+n-1}\left(a+b x+c x^2\right)^{p+1}}{c\, e^{n-1}(m+n+2 p+1)}+\frac{1}{c\, e^n (m+n+2 p+1)}\int (d+e x)^m \left(a+b x+c x^2\right)^p \cdot$$

$$\left(c\,e^n\,(m+n+2\,p+1)\,(f+g\,x)^n-c\,g^n\,(m+n+2\,p+1)\,(d+e\,x)^n+e\,g^n\,(m+p+n)\,(d+e\,x)^{n-2}\,(b\,d-2\,a\,e+(2\,c\,d-b\,e)\,x)\right)dx$$

Program code:

```
(* Int[(d_.+e_.*x_)^m_.*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  g^n*(d+e*x)^(m+n-1)*(a+b*x+c*x^2)^(p+1)/(c*e^(n-1)*(m+n+2*p+1)) +
  1/(c*e^n*(m+n+2*p+1))*Int[(d+e*x)^m*(a+b*x+c*x^2)^p*
    ExpandToSum[c*e^n*(m+n+2*p+1)*(f+g*x)^n-c*g^n*(m+n+2*p+1)*(d+e*x)^n+e*g^n*(m+p+n)*(d+e*x)^(n-2)*(b*d-2*a*e+(2*c*d-b*e)*x),x],x]
FreeQ[{a,b,c,d,e,f,g,m,p},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] &&
  NeQ[m+n+2*p+1,0] && IGtQ[n,0] *)
```

```
(* Int[(d_.+e_.*x_)^m_.*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  g^n*(d+e*x)^(m+n-1)*(a+c*x^2)^(p+1)/(c*e^(n-1)*(m+n+2*p+1)) +
  1/(c*e^n*(m+n+2*p+1))*Int[(d+e*x)^m*(a+c*x^2)^p*
    ExpandToSum[c*e^n*(m+n+2*p+1)*(f+g*x)^n-c*g^n*(m+n+2*p+1)*(d+e*x)^n-2*e*g^n*(m+p+n)*(d+e*x)^(n-2)*(a*e-c*d*x),x],x] /;
FreeQ[{a,c,d,e,f,g,m,p},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && NeQ[m+n+2*p+1,0] && IGtQ[n,0] *)
```

9: $\int (e\,x)^m\,(f+g\,x)^n\,\left(b\,x+c\,x^2\right)^p\,dx$ when $p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(e\,x)^m\,(b\,x+c\,x^2)^p}{x^{m+p}\,(b+c\,x)^p}==0$

Rule 1.2.1.4.2.2.9: If $p\notin\mathbb{Z}$, then

$$\int (e\,x)^m\,(f+g\,x)^n\,\left(b\,x+c\,x^2\right)^p\,dx \longrightarrow \frac{(e\,x)^m\,\left(b\,x+c\,x^2\right)^p}{x^{m+p}\,(b+c\,x)^p}\int x^{m+p}\,(f+g\,x)^n\,(b+c\,x)^p\,dx$$

Program code:

```
Int[(e_.*x_)^m_*(f_.+g_.*x_)^n_*(b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (e*x)^m*(b*x+c*x^2)^p/(x^(m+p)*(b+c*x)^p)*Int[x^(m+p)*(f+g*x)^n*(b+c*x)^p,x] /;
FreeQ[{b,c,e,f,g,m,n},x] && Not[IntegerQ[p]] && Not[IGtQ[n,0]]
```

10: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,c\,d^2+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,a>0\,\wedge\,d>0$

Derivation: Algebraic simplification

- **Basis: If** $c\,d^2+a\,e^2==0\,\wedge\,a>0\,\wedge\,d>0$, **then** $\left(a+c\,x^2\right)^p==\left(a-\frac{a\,e^2\,x^2}{d^2}\right)^p==(d+e\,x)^p\,\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p$

- **Rule 1.2.1.4.2.2.10: If** $e\,f-d\,g\neq 0\,\wedge\,c\,d^2+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,a>0\,\wedge\,d>0$, **then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+c\,x^2\right)^p\,dx\,\longrightarrow\,\int (d+e\,x)^{m+p}\,(f+g\,x)^n\,\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  Int[(d+e*x)^(m+p)*(f+g*x)^n*(a/d+c/e*x)^p,x] /;
FreeQ[{a,c,d,e,f,g,m,n},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && GtQ[a,0] && GtQ[d,0] && Not[IGtQ[m,0]] &&
```

11: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis: If** $c\,d^2-b\,d\,e+a\,e^2==0$, **then** $\partial_x\frac{\left(a+b\,x+c\,x^2\right)^p}{(d+e\,x)^p\,\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p}==0$

- **Basis: If** $c\,d^2-b\,d\,e+a\,e^2==0$, **then** $\frac{\left(a+b\,x+c\,x^2\right)^p}{(d+e\,x)^p\,\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p}==\frac{\left(a+b\,x+c\,x^2\right)^{\text{FracPart}[p]}}{(d+e\,x)^{\text{FracPart}[p]}\,\left(\frac{a}{d}+\frac{c\,x}{e}\right)^{\text{FracPart}[p]}}$

- **Note: This could replace the above rules in this section, but would result in slightly more complicated antiderivatives.**

- **Rule 1.2.1.4.2.2.11: If** $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2==0\,\wedge\,p\notin\mathbb{Z}$, **then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\frac{\left(a+b\,x+c\,x^2\right)^{\text{FracPart}[p]}}{(d+e\,x)^{\text{FracPart}[p]}\,\left(\frac{a}{d}+\frac{c\,x}{e}\right)^{\text{FracPart}[p]}}\int (d+e\,x)^{m+p}\,(f+g\,x)^n\,\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
(*(a+b*x+c*x^2)^p/((d+e*x)^p*(a*e+c*d*x)^p)*Int[(d+e*x)^(m+p)*(f+g*x)^n*(a*e+c*d*x)^p,x] /; *)
  (a+b*x+c*x^2)^FracPart[p]/((d+e*x)^FracPart[p]*(a/d+(c*x)/e)^FracPart[p])*Int[(d+e*x)^(m+p)*(f+g*x)^n*(a/d+c/e*x)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] && Not[IGtQ[m,0]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (a+c*x^2)^FracPart[p]/((d+e*x)^FracPart[p]*(a/d+(c*x)/e)^FracPart[p])*Int[(d+e*x)^(m+p)*(f+g*x)^n*(a/d+c/e*x)^p,x] /;
FreeQ[{a,c,d,e,f,g,m,n},x] && NeQ[e*f-d*g,0] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && Not[IGtQ[m,0]] && Not[IGtQ[n,0]]
```

3: $\int (d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p dx$ when $e f-d g\neq 0 \land b^2-4 a c\neq 0 \land c d^2-b d e+a e^2\neq 0 \land (m \mid n \mid p)\in \mathbb{Z}$

Derivation: Algebraic expansion

Rule 1.2.1.4.3: If $b^2-4 a c\neq 0 \land c d^2-b d e+a e^2\neq 0 \land (m \mid n \mid p)\in \mathbb{Z}$**, then**

$$\int (d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p dx \longrightarrow \int \text{ExpandIntegrand}\left[(d+e x)^m (f+g x)^n \left(a+b x+c x^2\right)^p, x\right] dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)^n*(a+b*x+c*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[p] &&
  (EqQ[p,1] && IntegersQ[m,n] || ILtQ[m,0] && ILtQ[n,0])
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)^n*(a+c*x^2)^p,x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && IntegerQ[p] &&
  (EqQ[p,1] && IntegersQ[m,n] || ILtQ[m,0] && ILtQ[n,0])
```

4: $\int \frac{(a+b x+c x^2)^p}{(d+e x)\,(f+g x)}\,dx$ when $e f-d g\neq 0\,\wedge\,b^2-4 a c\neq 0\,\wedge\,c d^2-b d e+a e^2\neq 0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,p>0$

Reference: Algebraic expansion

- **Basis:** $\frac{a+b x+c x^2}{d+e x}==\frac{(c d^2-b d e+a e^2)\,(f+g x)}{e\,(e f-d g)\,(d+e x)}-\frac{c d f-b e f+a e g-c\,(e f-d g)\,x}{e\,(e f-d g)}$

- **Rule 1.2.1.4.4: If** $e f-d g\neq 0\,\wedge\,b^2-4 a c\neq 0\,\wedge\,c d^2-b d e+a e^2\neq 0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,p>0$**, then**

$$\int \frac{(a+b x+c x^2)^p}{(d+e x)\,(f+g x)}\,dx\,\longrightarrow$$

$$\frac{c d^2-b d e+a e^2}{e\,(e f-d g)}\int \frac{(a+b x+c x^2)^{p-1}}{d+e x}\,dx-\frac{1}{e\,(e f-d g)}\int \frac{(c d f-b e f+a e g-c\,(e f-d g)\,x)\,(a+b x+c x^2)^{p-1}}{f+g x}\,dx$$

Program code:

```
Int[(a_.+b_.*x_+c_.*x_^2)^p_/((d_.+e_.*x_)*(f_.+g_.*x_)),x_Symbol] :=
  (c*d^2-b*d*e+a*e^2)/(e*(e*f-d*g))*Int[(a+b*x+c*x^2)^(p-1)/(d+e*x),x] -
  1/(e*(e*f-d*g))*Int[Simp[c*d*f-b*e*f+a*e*g-c*(e*f-d*g)*x,x]*(a+b*x+c*x^2)^(p-1)/(f+g*x),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && FractionQ[p] && GtQ[p,0]
```

```
Int[(a_+c_.*x_^2)^p_/((d_.+e_.*x_)*(f_.+g_.*x_)),x_Symbol] :=
  (c*d^2+a*e^2)/(e*(e*f-d*g))*Int[(a+c*x^2)^(p-1)/(d+e*x),x] -
  1/(e*(e*f-d*g))*Int[Simp[c*d*f+a*e*g-c*(e*f-d*g)*x,x]*(a+c*x^2)^(p-1)/(f+g*x),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && FractionQ[p] && GtQ[p,0]
```

5: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\land\,b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,(n\mid p)\in\mathbb{Z}\,\land\,m\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $q\in\mathbb{Z}^+$, then

$$(d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p = \frac{q}{e}\,\text{Subst}\left[x^{q\,(m+1)-1}\,\left(\frac{e\,f-d\,g}{e}+\frac{g\,x^q}{e}\right)^n\,\left(\frac{c\,d^2-b\,d\,e+a\,e^2}{e^2}-\frac{(2\,c\,d-b\,e)\,x^q}{e^2}+\frac{c\,x^{2\,q}}{e^2}\right)^p,\,x,\,(d+e\,x)^{1/q}\right]\,\partial_x\,(d+e\,x)^{1/q}$$

Rule 1.2.1.4.5: If $e\,f-d\,g\neq 0\,\land\,b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,(n\mid p)\in\mathbb{Z}\,\land\,m\in\mathbb{F}$, let $q=\text{Denominator}[m]$, then

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\frac{q}{e}\,\text{Subst}\left[\int x^{q\,(m+1)-1}\,\left(\frac{e\,f-d\,g}{e}+\frac{g\,x^q}{e}\right)^n\,\left(\frac{c\,d^2-b\,d\,e+a\,e^2}{e^2}-\frac{(2\,c\,d-b\,e)\,x^q}{e^2}+\frac{c\,x^{2\,q}}{e^2}\right)^p\,dx,\,x,\,(d+e\,x)^{1/q}\right]$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  With[{q=Denominator[m]},
  q/e*Subst[Int[x^(q*(m+1)-1)*((e*f-d*g)/e+g*x^q/e)^n*
    ((c*d^2-b*d*e+a*e^2)/e^2-(2*c*d-b*e)*x^q/e^2+c*x^(2*q)/e^2)^p,x],x,(d+e*x)^(1/q)]] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegersQ[n,p] && FractionQ[m]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  With[{q=Denominator[m]},
  q/e*Subst[Int[x^(q*(m+1)-1)*((e*f-d*g)/e+g*x^q/e)^n*((c*d^2+a*e^2)/e^2-2*c*d*x^q/e^2+c*x^(2*q)/e^2)^p,x],x,(d+e*x)^(1/q)]] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && IntegersQ[n,p] && FractionQ[m]
```

6. $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,m-n=0\,\wedge\,e\,f+d\,g=0$

1: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $m-n=0\,\wedge\,e\,f+d\,g=0\,\wedge\,(m\in\mathbb{Z}\,\vee\,d>0\,\wedge\,f>0)$

Derivation: Algebraic simplification

Basis: If $e\,f+d\,g=0\,\wedge\,d>0\,\wedge\,f>0$**, then** $(d+e\,x)^m\,(f+g\,x)^m=\left(d\,f+e\,g\,x^2\right)^m$

Rule 1.2.1.4.6.1: If $m-n=0\,\wedge\,e\,f+d\,g=0\,\wedge\,(m\in\mathbb{Z}\,\vee\,d>0\,\wedge\,f>0)$**, then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\int \left(d\,f+e\,g\,x^2\right)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[(d*f+e*g*x^2)^m*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && EqQ[m-n,0] && EqQ[e*f+d*g,0] && (IntegerQ[m] || GtQ[d,0] && GtQ[f,0])
```

```
Int[(d_+e_.*x_)^m_*(f_+g_.*x_)^n_*(a_.+c_.*x_^2)^p_.,x_Symbol] :=
  Int[(d*f+e*g*x^2)^m*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,f,g,m,n,p},x] && EqQ[m-n,0] && EqQ[e*f+d*g,0] && (IntegerQ[m] || GtQ[d,0] && GtQ[f,0])
```

2: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $m-n=0\,\wedge\,e\,f+d\,g=0$

Derivation: Piecewise constant extraction

Basis: If $e\,f+d\,g=0$**, then** $\partial_x \frac{(d+e\,x)^m\,(f+g\,x)^m}{\left(d\,f+e\,g\,x^2\right)^m}=0$

Rule 1.2.1.4.6.2: If $m-n=0\,\wedge\,e\,f+d\,g=0$**, then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\frac{(d+e\,x)^{\text{FracPart}[m]}\,(f+g\,x)^{\text{FracPart}[m]}}{\left(d\,f+e\,g\,x^2\right)^{\text{FracPart}[m]}}\int \left(d\,f+e\,g\,x^2\right)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  (d+e*x)^FracPart[m]*(f+g*x)^FracPart[m]/(d*f+e*g*x^2)^FracPart[m]*Int[(d*f+e*g*x^2)^m*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && EqQ[m-n,0] && EqQ[e*f+d*g,0]
```

```
Int[(d_+e_.*x_)^m_*(f_+g_.*x_)^n_*(a_.+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^FracPart[m]*(f+g*x)^FracPart[m]/(d*f+e*g*x^2)^FracPart[m]*Int[(d*f+e*g*x^2)^m*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,f,g,m,n,p},x] && EqQ[m-n,0] && EqQ[e*f+d*g,0]
```

7. $\int \frac{(d+e\,x)^m\,(f+g\,x)^n}{a+b\,x+c\,x^2}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,m\notin\mathbb{Z}\,\wedge\,n\notin\mathbb{Z}$

  1. $\int \frac{(d+e\,x)^m\,(f+g\,x)^n}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,m\notin\mathbb{Z}\,\wedge\,n\notin\mathbb{Z}\,\wedge\,m>0$

    1. $\int \frac{(d+e\,x)^m\,(f+g\,x)^n}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,m\notin\mathbb{Z}\,\wedge\,n\notin\mathbb{Z}\,\wedge\,m>0\,\wedge\,n>0$

      1: $\int \frac{(d+e\,x)^m\,(f+g\,x)^n}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,m\notin\mathbb{Z}\,\wedge\,n\notin\mathbb{Z}\,\wedge\,m>0\,\wedge\,n>1$

Reference: Algebraic expansion

- **Basis:** $\frac{(d+e\,x)^m\,(f+g\,x)^n}{a+b\,x+c\,x^2}==\frac{g\,(2\,c\,e\,f+c\,d\,g-b\,e\,g+c\,e\,g\,x)\,(d+e\,x)^{m-1}\,(f+g\,x)^{n-2}}{c^2}+\frac{1}{c^2\,(a+b\,x+c\,x^2)}$
$\left(c^2\,d\,f^2-2\,a\,c\,e\,f\,g-a\,c\,d\,g^2+a\,b\,e\,g^2+\left(c^2\,e\,f^2+2\,c^2\,d\,f\,g-2\,b\,c\,e\,f\,g-b\,c\,d\,g^2+b^2\,e\,g^2-a\,c\,e\,g^2\right)x\right)\,(d+e\,x)^{m-1}\,(f+g\,x)^{n-2}$

Rule 1.2.1.4.7.1.1.1: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,m\notin\mathbb{Z}\,\wedge\,n\notin\mathbb{Z}\,\wedge\,m>0\,\wedge\,n>1$**, then**

$$\int \frac{(d+e\,x)^m\,(f+g\,x)^n}{a+b\,x+c\,x^2}\,dx\;\longrightarrow$$
$$\frac{g}{c^2}\int (2\,c\,e\,f+c\,d\,g-b\,e\,g+c\,e\,g\,x)\,(d+e\,x)^{m-1}\,(f+g\,x)^{n-2}\,dx+$$
$$\frac{1}{c^2}\int \frac{1}{a+b\,x+c\,x^2}\left(c^2\,d\,f^2-2\,a\,c\,e\,f\,g-a\,c\,d\,g^2+a\,b\,e\,g^2+\left(c^2\,e\,f^2+2\,c^2\,d\,f\,g-2\,b\,c\,e\,f\,g-b\,c\,d\,g^2+b^2\,e\,g^2-a\,c\,e\,g^2\right)x\right)\,(d+e\,x)^{m-1}\,(f+g\,x)^{n-2}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  g/c^2*Int[Simp[2*c*e*f+c*d*g-b*e*g+c*e*g*x,x]*(d+e*x)^(m-1)*(f+g*x)^(n-2),x] +
  1/c^2*Int[Simp[c^2*d*f^2-2*a*c*e*f*g-a*c*d*g^2+a*b*e*g^2+(c^2*e*f^2+2*c^2*d*f*g-2*b*c*e*f*g-b*c*d*g^2+b^2*e*g^2-a*c*e*g^2)*x,x]*
    (d+e*x)^(m-1)*(f+g*x)^(n-2)/(a+b*x+c*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && GtQ[m,0] && GtQ
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_/(a_+c_.*x_^2),x_Symbol] :=
  g/c*Int[Simp[2*e*f+d*g+e*g*x,x]*(d+e*x)^(m-1)*(f+g*x)^(n-2),x] +
  1/c*Int[Simp[c*d*f^2-2*a*e*f*g-a*d*g^2+(c*e*f^2+2*c*d*f*g-a*e*g^2)*x,x]*(d+e*x)^(m-1)*(f+g*x)^(n-2)/(a+c*x^2),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && GtQ[m,0] && GtQ[n,1]
```

2: $\int \frac{(d+e\,x)^m\,(f+g\,x)^n}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}\,\land\,m>0\,\land\,n>0$

- **Reference: Algebraic expansion**

- **Basis:** $\frac{(d+e\,x)^m\,(f+g\,x)^n}{a+b\,x+c\,x^2}==\frac{e\,g\,(d+e\,x)^{m-1}\,(f+g\,x)^{n-1}}{c}+\frac{(c\,d\,f-a\,e\,g+(c\,e\,f+c\,d\,g-b\,e\,g)\,x)\,(d+e\,x)^{m-1}\,(f+g\,x)^{n-1}}{c\,(a+b\,x+c\,x^2)}$

- **Rule 1.2.1.4.7.1.1.2: If** $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}\,\land\,m>0\,\land\,n>0$**, then**

$$\int \frac{(d+e\,x)^m\,(f+g\,x)^n}{a+b\,x+c\,x^2}\,dx\,\longrightarrow$$
$$\frac{e\,g}{c}\int (d+e\,x)^{m-1}\,(f+g\,x)^{n-1}\,dx+\frac{1}{c}\int \frac{(c\,d\,f-a\,e\,g+(c\,e\,f+c\,d\,g-b\,e\,g)\,x)\,(d+e\,x)^{m-1}\,(f+g\,x)^{n-1}}{a+b\,x+c\,x^2}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  e*g/c*Int[(d+e*x)^(m-1)*(f+g*x)^(n-1),x] +
  1/c*Int[Simp[c*d*f-a*e*g+(c*e*f+c*d*g-b*e*g)*x,x]*(d+e*x)^(m-1)*(f+g*x)^(n-1)/(a+b*x+c*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IntegerQ[m]] && Not[IntegerQ[n]] && GtQ[m,0] && GtQ[n,0]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_/(a_+c_.*x_^2),x_Symbol] :=
  e*g/c*Int[(d+e*x)^(m-1)*(f+g*x)^(n-1),x] +
  1/c*Int[Simp[c*d*f-a*e*g+(c*e*f+c*d*g)*x,x]*(d+e*x)^(m-1)*(f+g*x)^(n-1)/(a+c*x^2),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && GtQ[m,0] && GtQ[n,0]
```

2: $\int \frac{(d+e x)^m (f+g x)^n}{a+b x+c x^2} dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land m \notin \mathbb{Z} \land n \notin \mathbb{Z} \land m>0 \land n<-1$

Reference: Algebraic expansion

- **Basis:** $\frac{(d+e x)^m (f+g x)^n}{a+b x+c x^2} = -\frac{g (e f-d g) (d+e x)^{m-1} (f+g x)^n}{c f^2-b f g+a g^2}+\frac{(c d f-b d g+a e g+c (e f-d g) x) (d+e x)^{m-1} (f+g x)^{n+1}}{(c f^2-b f g+a g^2) (a+b x+c x^2)}$

Rule 1.2.1.4.7.1.2: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land m \notin \mathbb{Z} \land n \notin \mathbb{Z} \land m>0 \land n<-1$**, then**

$$\int \frac{(d+e x)^m (f+g x)^n}{a+b x+c x^2} dx \longrightarrow$$

$$-\frac{g (e f-d g)}{c f^2-b f g+a g^2} \int (d+e x)^{m-1} (f+g x)^n dx+\frac{1}{c f^2-b f g+a g^2} \int \frac{(c d f-b d g+a e g+c (e f-d g) x) (d+e x)^{m-1} (f+g x)^{n+1}}{a+b x+c x^2} dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  -g*(e*f-d*g)/(c*f^2-b*f*g+a*g^2)*Int[(d+e*x)^(m-1)*(f+g*x)^n,x] +
  1/(c*f^2-b*f*g+a*g^2)*
    Int[Simp[c*d*f-b*d*g+a*e*g+c*(e*f-d*g)*x,x]*(d+e*x)^(m-1)*(f+g*x)^(n+1)/(a+b*x+c*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IntegerQ[m]] && Not[IntegerQ[n]] && GtQ[m,0] && LtQ[n,-1]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_/(a_+c_.*x_^2),x_Symbol] :=
  -g*(e*f-d*g)/(c*f^2+a*g^2)*Int[(d+e*x)^(m-1)*(f+g*x)^n,x] +
  1/(c*f^2+a*g^2)*
    Int[Simp[c*d*f+a*e*g+c*(e*f-d*g)*x,x]*(d+e*x)^(m-1)*(f+g*x)^(n+1)/(a+c*x^2),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && GtQ[m,0] && LtQ[n,-1]
```

2. $\int \frac{(d+e x)^m (f+g x)^n}{a+b x+c x^2} dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land m \notin \mathbb{Z} \land n \notin \mathbb{Z}$

1: $\int \frac{(d+e x)^m}{\sqrt{f+g x} (a+b x+c x^2)} dx$ when $b^2-4 a c \neq 0 \bigwedge c d^2-b d e+a e^2 \neq 0 \bigwedge m+\frac{1}{2} \in \mathbb{Z}^+$

Derivation: Algebraic expansion

- **Basis: If** $q \to \sqrt{b^2-4 a c}$**, then** $\frac{d+e x}{a+b x+c x^2} = \frac{2 c d-e (b-q)}{q (b-q+2 c x)}-\frac{2 c d-e (b+q)}{q (b+q+2 c x)}$
- **Rule 1.2.1.4.7.2.1: If** $b^2-4 a c \neq 0 \bigwedge c d^2-b d e+a e^2 \neq 0 \bigwedge m+\frac{1}{2} \in \mathbb{Z}^+$**, then**

$$\int \frac{(d+e\,x)^m}{\sqrt{f+g\,x}\,\left(a+b\,x+c\,x^2\right)}\,dx\;\longrightarrow\;\int \frac{1}{\sqrt{d+e\,x}\,\sqrt{f+g\,x}}\,\text{ExpandIntegrand}\left[\frac{(d+e\,x)^{m+\frac{1}{2}}}{a+b\,x+c\,x^2},\,x\right]dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_/(Sqrt[f_.+g_.*x_]*(a_.+b_.*x_+c_.*x_^2)),x_Symbol] :=
  Int[ExpandIntegrand[1/(Sqrt[d+e*x]*Sqrt[f+g*x]),(d+e*x)^(m+1/2)/(a+b*x+c*x^2),x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IGtQ[m+1/2,0]
```

```
Int[(d_.+e_.*x_)^m_/(Sqrt[f_.+g_.*x_]*(a_.+c_.*x_^2)),x_Symbol] :=
  Int[ExpandIntegrand[1/(Sqrt[d+e*x]*Sqrt[f+g*x]),(d+e*x)^(m+1/2)/(a+c*x^2),x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] && IGtQ[m+1/2,0]
```

2: $\int \frac{(d+e\,x)^m\,(f+g\,x)^n}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}$

Derivation: Algebraic expansion

- **Basis: If** $q\to\sqrt{b^2-4\,a\,c}$ **, then** $\frac{1}{a+b\,z+c\,z^2}==\frac{2\,c}{q\,(b-q+2\,c\,z)}-\frac{2\,c}{q\,(b+q+2\,c\,z)}$
- **Rule 1.2.1.4.7.2.2: If** $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}$**, then**

$$\int \frac{(d+e\,x)^m\,(f+g\,x)^n}{a+b\,x+c\,x^2}\,dx\;\longrightarrow\;\int (d+e\,x)^m\,(f+g\,x)^n\,\text{ExpandIntegrand}\left[\frac{1}{a+b\,x+c\,x^2},\,x\right]dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)^n,1/(a+b*x+c*x^2),x],x] /;
FreeQ[{a,b,c,d,e,f,g,m,n},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[m]] && Not[IntegerQ[n]]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_/(a_+c_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)^n,1/(a+c*x^2),x],x] /;
FreeQ[{a,c,d,e,f,g,m,n},x] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[m]] && Not[IntegerQ[n]]
```

8: $\int x^2 (d+e x)^m \left(a+b x+c x^2\right)^p \, dx$ when $b e (m+p+2)+2 c d (p+1)=0 \land b d (p+1)+a e (m+1)=0 \land m+2 p+3\neq 0$

Derivation: Special case of one step of the Ostrogradskiy-Hermite integration method

Rule 1.2.1.4.8: If $b e (m+p+2)+2 c d (p+1)=0 \land b d (p+1)+a e (m+1)=0 \land m+2 p+3\neq 0$, then

$$\int x^2 (d+e x)^m \left(a+b x+c x^2\right)^p \, dx \longrightarrow \frac{(d+e x)^{m+1} \left(a+b x+c x^2\right)^{p+1}}{c e (m+2 p+3)}$$

Program code:

```
Int[x_^2*(d_.+e_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)/(c*e*(m+2*p+3)) /;
FreeQ[{a,b,c,d,e,m,p},x] && EqQ[b*e*(m+p+2)+2*c*d*(p+1),0] && EqQ[b*d*(p+1)+a*e*(m+1),0] && NeQ[m+2*p+3,0]
```

```
Int[x_^2*(d_.+e_.*x_)^m_.*(a_.+c_.*x_^2)^p_.,x_Symbol] :=
  (d+e*x)^(m+1)*(a+c*x^2)^(p+1)/(c*e*(m+2*p+3)) /;
FreeQ[{a,c,d,e,m,p},x] && EqQ[d*(p+1),0] && EqQ[a*(m+1),0] && NeQ[m+2*p+3,0]
```

9: $\int (g x)^n (d+e x)^m \left(a+b x+c x^2\right)^p \, dx$ when $b^2-4 a c\neq 0 \land c d^2-b d e+a e^2\neq 0 \land m-p=0 \land b d+a e=0 \land c d+b e=0$

Derivation: Piecewise constant extraction

- **Basis: If $b d+a e=0 \land c d+b e=0$, then $\partial_x \frac{(d+e x)^p \left(a+b x+c x^2\right)^p}{\left(a d+c e x^3\right)^p}=0$**

Rule 1.2.1.4.9: If $m-p=0 \land b d+a e=0 \land c d+b e=0$, then

$$\int (g x)^n (d+e x)^m \left(a+b x+c x^2\right)^p \, dx \longrightarrow \frac{(d+e x)^{\text{FracPart}[p]} \left(a+b x+c x^2\right)^{\text{FracPart}[p]}}{\left(a d+c e x^3\right)^{\text{FracPart}[p]}} \int (g x)^n \left(a d+c e x^3\right)^p \, dx$$

Program code:

```
Int[(g_.*x_)^n_*(d_.+e_.*x_)^m_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^FracPart[p]*(a+b*x+c*x^2)^FracPart[p]/(a*d+c*e*x^3)^FracPart[p]*Int[(g*x)^n*(a*d+c*e*x^3)^p,x] /;
FreeQ[{a,b,c,d,e,g,m,n,p},x] && EqQ[m-p,0] && EqQ[b*d+a*e,0] && EqQ[c*d+b*e,0]
```

10. $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\bigwedge\,b^2-4\,a\,c\neq 0\,\bigwedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\bigwedge\,2\,m\in\mathbb{Z}\,\bigwedge\,n^2==\frac{1}{4}\,\bigwedge\,p^2==\frac{1}{4}$

1. $\int (d+e\,x)^m\,(f+g\,x)^n\,\sqrt{a+b\,x+c\,x^2}\,dx$ when $e\,f-d\,g\neq 0\,\bigwedge\,b^2-4\,a\,c\neq 0\,\bigwedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\bigwedge\,2\,m\in\mathbb{Z}\,\bigwedge\,n^2==\frac{1}{4}$

1. $\int (d+e\,x)^m\,\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,m\in\mathbb{Z}$

1: $\int (d+e\,x)^m\,\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,m\in\mathbb{Z}\,\wedge\,m<-1$

Derivation: Integration by parts

- **Basis:** $\partial_x\left(\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}\right)==\frac{b\,f+a\,g+2\,(c\,f+b\,g)\,x+3\,c\,g\,x^2}{2\,\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}$

- **Rule 1.2.1.4.10.1.1.1: If** $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,m\in\mathbb{Z}\,\wedge\,m<-1$**, then**

$$\int (d+e\,x)^m\,\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}\,dx\,\longrightarrow$$

$$\frac{(d+e\,x)^{m+1}\,\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}{e\,(m+1)}-\frac{1}{2\,e\,(m+1)}\int\frac{(d+e\,x)^{m+1}\,\left(b\,f+a\,g+2\,(c\,f+b\,g)\,x+3\,c\,g\,x^2\right)}{\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Sqrt[f_.+g_.*x_]*Sqrt[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2]/(e*(m+1)) -
  1/(2*e*(m+1))*Int[(d+e*x)^(m+1)/(Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2])*Simp[b*f+a*g+2*(c*f+b*g)*x+3*c*g*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[2*m] && LtQ[m,-1]
```

```
Int[(d_.+e_.*x_)^m_.*Sqrt[f_.+g_.*x_]*Sqrt[a_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Sqrt[f+g*x]*Sqrt[a+c*x^2]/(e*(m+1)) -
  1/(2*e*(m+1))*Int[(d+e*x)^(m+1)/(Sqrt[f+g*x]*Sqrt[a+c*x^2])*Simp[a*g+2*c*f*x+3*c*g*x^2,x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && IntegerQ[2*m] && LtQ[m,-1]
```

2: $\int (d+e\,x)^m\,\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,m\in\mathbb{Z}\,\wedge\,m\nless -1$

- **Rule 1.2.1.4.10.1.1.2: If** $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,m\in\mathbb{Z}\,\wedge\,m\nless -1$**, then**

$$\int (d+e x)^m \sqrt{f+g x}\, \sqrt{a+b x+c x^2}\, dx \longrightarrow$$

$$\frac{2 (d+e x)^{m+1} \sqrt{f+g x}\, \sqrt{a+b x+c x^2}}{e (2 m+5)} -$$
$$\frac{1}{e (2 m+5)} \int \left((d+e x)^m \left(b d f-3 a e f+a d g+2 (c d f-b e f+b d g-a e g) x-(c e f-3 c d g+b e g) x^2\right)\right) \Big/ \left(\sqrt{f+g x}\, \sqrt{a+b x+c x^2}\right) dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Sqrt[f_.+g_.*x_]*Sqrt[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  2*(d+e*x)^(m+1)*Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2]/(e*(2*m+5)) -
  1/(e*(2*m+5))*Int[(d+e*x)^m/(Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2])*
     Simp[b*d*f-3*a*e*f+a*d*g+2*(c*d*f-b*e*f+b*d*g-a*e*g)*x-(c*e*f-3*c*d*g+b*e*g)*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,g,m},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[2*m] && Not[LtQ[m,-1]]
```

```
Int[(d_.+e_.*x_)^m_.*Sqrt[f_.+g_.*x_]*Sqrt[a_+c_.*x_^2],x_Symbol] :=
  2*(d+e*x)^(m+1)*Sqrt[f+g*x]*Sqrt[a+c*x^2]/(e*(2*m+5)) +
  1/(e*(2*m+5))*Int[(d+e*x)^m/(Sqrt[f+g*x]*Sqrt[a+c*x^2])*
     Simp[3*a*e*f-a*d*g-2*(c*d*f-a*e*g)*x+(c*e*f-3*c*d*g)*x^2,x],x] /;
FreeQ[{a,c,d,e,f,g,m},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && IntegerQ[2*m] && Not[LtQ[m,-1]]
```

2. $\int \frac{(d+e x)^m \sqrt{a+b x+c x^2}}{\sqrt{f+g x}}\, dx$ when $e f-d g \neq 0 \wedge b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge 2 m \in \mathbb{Z}$

1: $\int \frac{(d+e x)^m \sqrt{a+b x+c x^2}}{\sqrt{f+g x}}\, dx$ when $e f-d g \neq 0 \wedge b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge 2 m \in \mathbb{Z} \wedge m>0$

Rule 1.2.1.4.10.1.2.1: If $e f-d g \neq 0 \wedge b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge 2 m \in \mathbb{Z} \wedge m>0$, then

$$\int \frac{(d+e x)^m \sqrt{a+b x+c x^2}}{\sqrt{f+g x}}\, dx \longrightarrow$$

$$\frac{2 (d+e x)^m \sqrt{f+g x}\, \sqrt{a+b x+c x^2}}{g (2 m+3)} - \frac{1}{g (2 m+3)} \int \frac{(d+e x)^{m-1}}{\sqrt{f+g x}\, \sqrt{a+b x+c x^2}} \cdot$$

$$\left(b\,d\,f+2\,a\,(e\,f\,m-d\,g\,(m+1))+(2\,c\,d\,f-2\,a\,e\,g+b\,(e\,f-d\,g)\,(2\,m+1))\,x-(b\,e\,g+2\,c\,(d\,g\,m-e\,f\,(m+1)))\,x^2\right)\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_.*Sqrt[a_.+b_.*x_+c_.*x_^2]/Sqrt[f_.+g_.*x_],x_Symbol] :=
  2*(d+e*x)^m*Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2]/(g*(2*m+3)) -
  1/(g*(2*m+3))*Int[(d+e*x)^(m-1)/(Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2])*
    Simp[b*d*f+2*a*(e*f*m-d*g*(m+1))+(2*c*d*f-2*a*e*g+b*(e*f-d*g)*(2*m+1))*x-(b*e*g+2*c*(d*g*m-e*f*(m+1)))*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[2*m] && GtQ[m,0]
```

```
Int[(d_.+e_.*x_)^m_.*Sqrt[a_+c_.*x_^2]/Sqrt[f_.+g_.*x_],x_Symbol] :=
  2*(d+e*x)^m*Sqrt[f+g*x]*Sqrt[a+c*x^2]/(g*(2*m+3)) -
  1/(g*(2*m+3))*Int[(d+e*x)^(m-1)/(Sqrt[f+g*x]*Sqrt[a+c*x^2])*
    Simp[2*a*(e*f*m-d*g*(m+1))+(2*c*d*f-2*a*e*g)*x-(2*c*(d*g*m-e*f*(m+1)))*x^2,x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && IntegerQ[2*m] && GtQ[m,0]
```

2. $\int \frac{(d+e\,x)^m\,\sqrt{a+b\,x+c\,x^2}}{\sqrt{f+g\,x}}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,m\in\mathbb{Z}\,\wedge\,m<0$

1: $\int \frac{\sqrt{a+b\,x+c\,x^2}}{(d+e\,x)\,\sqrt{f+g\,x}}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0$

Derivation: Algebraic expansion

- **Basis:** $\frac{\sqrt{a+b\,x+c\,x^2}}{d+e\,x}==\frac{c\,d^2-b\,d\,e+a\,e^2}{e^2\,(d+e\,x)\,\sqrt{a+b\,x+c\,x^2}}-\frac{c\,d-b\,e-c\,e\,x}{e^2\,\sqrt{a+b\,x+c\,x^2}}$

- **Rule 1.2.1.4.10.1.2.2.1: If** $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0$**, then**

$$\int \frac{\sqrt{a+b\,x+c\,x^2}}{(d+e\,x)\,\sqrt{f+g\,x}}\,dx\,\longrightarrow\,\frac{c\,d^2-b\,d\,e+a\,e^2}{e^2}\int \frac{1}{(d+e\,x)\,\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx-\frac{1}{e^2}\int \frac{c\,d-b\,e-c\,e\,x}{\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$$

Program code:

```
Int[Sqrt[a_.+b_.*x_+c_.*x_^2]/((d_.+e_.*x_)*Sqrt[f_.+g_.*x_]),x_Symbol] :=
  (c*d^2-b*d*e+a*e^2)/e^2*Int[1/((d+e*x)*Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2]),x] -
  1/e^2*Int[(c*d-b*e-c*e*x)/(Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2]),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[Sqrt[a_+c_.*x_^2]/((d_.+e_.*x_)*Sqrt[f_.+g_.*x_]),x_Symbol] :=
  (c*d^2+a*e^2)/e^2*Int[1/((d+e*x)*Sqrt[f+g*x]*Sqrt[a+c*x^2]),x] -
  1/e^2*Int[(c*d-c*e*x)/(Sqrt[f+g*x]*Sqrt[a+c*x^2]),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0]
```

2: $\int \frac{(d+e x)^m \sqrt{a+b x+c x^2}}{\sqrt{f+g x}} \, dx$ when $e f-d g \neq 0 \land b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 m \in \mathbb{Z} \land m<-1$

Rule 1.2.1.4.10.1.2.2.2: If $e f-d g \neq 0 \land b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 m \in \mathbb{Z} \land m<-1$, **then**

$$\int \frac{(d+e x)^m \sqrt{a+b x+c x^2}}{\sqrt{f+g x}} \, dx \longrightarrow$$

$$\frac{(d+e x)^{m+1} \sqrt{f+g x} \sqrt{a+b x+c x^2}}{(m+1)(e f-d g)}-\frac{1}{2(m+1)(e f-d g)} \int \frac{(d+e x)^{m+1} \left(b f+a g(2 m+3)+2(c f+b g(m+2)) x+c g(2 m+5) x^2\right)}{\sqrt{f+g x} \sqrt{a+b x+c x^2}} \, dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Sqrt[a_.+b_.*x_+c_.*x_^2]/Sqrt[f_.+g_.*x_],x_Symbol] :=
  (d+e*x)^(m+1)*Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2]/((m+1)*(e*f-d*g)) -
  1/(2*(m+1)*(e*f-d*g))*Int[(d+e*x)^(m+1)/(Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2])*
    Simp[b*f+a*g*(2*m+3)+2*(c*f+b*g*(m+2))*x+c*g*(2*m+5)*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[2*m] && LtQ[m,-1]
```

```
Int[(d_.+e_.*x_)^m_.*Sqrt[a_+c_.*x_^2]/Sqrt[f_.+g_.*x_],x_Symbol] :=
  (d+e*x)^(m+1)*Sqrt[f+g*x]*Sqrt[a+c*x^2]/((m+1)*(e*f-d*g)) -
  1/(2*(m+1)*(e*f-d*g))*Int[(d+e*x)^(m+1)/(Sqrt[f+g*x]*Sqrt[a+c*x^2])*
    Simp[a*g*(2*m+3)+2*(c*f)*x+c*g*(2*m+5)*x^2,x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && IntegerQ[2*m] && LtQ[m,-1]
```

2. $\int \frac{(d+e x)^m (f+g x)^n}{\sqrt{a+b x+c x^2}} \, dx$ when $e f-d g \neq 0 \bigwedge b^2-4 a c \neq 0 \bigwedge c d^2-b d e+a e^2 \neq 0 \bigwedge 2 m \in \mathbb{Z} \bigwedge n^2 == \frac{1}{4}$

1. $\int \frac{(d+e x)^m}{\sqrt{f+g x} \sqrt{a+b x+c x^2}} \, dx$ when $e f-d g \neq 0 \land b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 m \in \mathbb{Z}$

1. $\int \frac{(d+e\,x)^m}{\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,m\in\mathbb{Z}\,\wedge\,m>0$

1: $\int \frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0$

Derivation: Piecewise constant extraction and integration by substitution

- **Rule 1.2.1.4.10.2.1.1.1: If** $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0$**, let** $q\to\sqrt{b^2-4\,a\,c}$**, then**

$$\int \frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx\,\longrightarrow$$

$$\left(\sqrt{2}\,\sqrt{2\,c\,f-g\,(b+q)}\,\sqrt{b-q+2\,c\,x}\,(d+e\,x)\,\sqrt{\frac{(e\,f-d\,g)\,(b+q+2\,c\,x)}{(2\,c\,f-g\,(b+q))\,(d+e\,x)}}\,\sqrt{\frac{(e\,f-d\,g)\,(2\,a+(b+q)\,x)}{(b\,f+q\,f-2\,a\,g)\,(d+e\,x)}}\right)\Big/$$
$$\left(g\,\sqrt{2\,c\,d-e\,(b+q)}\,\sqrt{\frac{2\,a\,c}{b+q}+c\,x}\,\sqrt{a+b\,x+c\,x^2}\right)\cdot$$
$$\text{EllipticPi}\left[\frac{e\,(2\,c\,f-g\,(b+q))}{g\,(2\,c\,d-e\,(b+q))},\,\text{ArcSin}\left[\frac{\sqrt{2\,c\,d-e\,(b+q)}\,\sqrt{f+g\,x}}{\sqrt{2\,c\,f-g\,(b+q)}\,\sqrt{d+e\,x}}\right],\,\frac{(b\,d+q\,d-2\,a\,e)\,(2\,c\,f-g\,(b+q))}{(b\,f+q\,f-2\,a\,g)\,(2\,c\,d-e\,(b+q))}\right]$$

Program code:

```
Int[Sqrt[d_.+e_.*x_]/(Sqrt[f_.+g_.*x_]*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  Sqrt[2]*Sqrt[2*c*f-g*(b+q)]*Sqrt[b-q+2*c*x]*(d+e*x)*
    Sqrt[(e*f-d*g)*(b+q+2*c*x)/((2*c*f-g*(b+q))*(d+e*x))]*
    Sqrt[(e*f-d*g)*(2*a+(b+q)*x)/((b*f+q*f-2*a*g)*(d+e*x))]/
   (g*Sqrt[2*c*d-e*(b+q)]*Sqrt[2*a*c/(b+q)+c*x]*Sqrt[a+b*x+c*x^2])*
    EllipticPi[e*(2*c*f-g*(b+q))/(g*(2*c*d-e*(b+q))),
      ArcSin[Sqrt[2*c*d-e*(b+q)]*Sqrt[f+g*x]/(Sqrt[2*c*f-g*(b+q)]*Sqrt[d+e*x])],
      (b*d+q*d-2*a*e)*(2*c*f-g*(b+q))/((b*f+q*f-2*a*g)*(2*c*d-e*(b+q)))]] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[Sqrt[d_.+e_.*x_]/(Sqrt[f_.+g_.*x_]*Sqrt[a_+c_.*x_^2]),x_Symbol] :=
  With[{q=Rt[-4*a*c,2]},
  Sqrt[2]*Sqrt[2*c*f-g*q]*Sqrt[-q+2*c*x]*(d+e*x)*
    Sqrt[(e*f-d*g)*(q+2*c*x)/((2*c*f-g*q)*(d+e*x))]*
    Sqrt[(e*f-d*g)*(2*a+q*x)/((q*f-2*a*g)*(d+e*x))]/
   (g*Sqrt[2*c*d-e*q]*Sqrt[2*a*c/q+c*x]*Sqrt[a+c*x^2])*
    EllipticPi[e*(2*c*f-g*q)/(g*(2*c*d-e*q)),
      ArcSin[Sqrt[2*c*d-e*q]*Sqrt[f+g*x]/(Sqrt[2*c*f-g*q]*Sqrt[d+e*x])],
      (q*d-2*a*e)*(2*c*f-g*q)/((q*f-2*a*g)*(2*c*d-e*q))]] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0]
```

2: $\int \frac{(d+e x)^{3/2}}{\sqrt{f+g x}\,\sqrt{a+b x+c x^2}}\,dx$ when $e f-d g\neq 0\land b^2-4 a c\neq 0\land c d^2-b d e+a e^2\neq 0$

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{(d+e x)^{3/2}}{\sqrt{f+g x}}==\frac{e\sqrt{d+e x}\,\sqrt{f+g x}}{g}-\frac{(e f-d g)\sqrt{d+e x}}{g\sqrt{f+g x}}$

- **Rule 1.2.1.4.10.2.1.1.2: If** $e f-d g\neq 0\land b^2-4 a c\neq 0\land c d^2-b d e+a e^2\neq 0$**, then**

$$\int \frac{(d+e x)^{3/2}}{\sqrt{f+g x}\,\sqrt{a+b x+c x^2}}\,dx\rightarrow \frac{e}{g}\int \frac{\sqrt{d+e x}\,\sqrt{f+g x}}{\sqrt{a+b x+c x^2}}\,dx-\frac{(e f-d g)}{g}\int \frac{\sqrt{d+e x}}{\sqrt{f+g x}\,\sqrt{a+b x+c x^2}}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^(3/2)/(Sqrt[f_.+g_.*x_]*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  e/g*Int[Sqrt[d+e*x]*Sqrt[f+g*x]/Sqrt[a+b*x+c*x^2],x] -
  (e*f-d*g)/g*Int[Sqrt[d+e*x]/(Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2]),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[(d_.+e_.*x_)^(3/2)/(Sqrt[f_.+g_.*x_]*Sqrt[a_+c_.*x_^2]),x_Symbol] :=
  e/g*Int[Sqrt[d+e*x]*Sqrt[f+g*x]/Sqrt[a+c*x^2],x] -
  (e*f-d*g)/g*Int[Sqrt[d+e*x]/(Sqrt[f+g*x]*Sqrt[a+c*x^2]),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0]
```

3: $\int \frac{(d+e\,x)^m}{\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,m\in\mathbb{Z}\,\wedge\,m\geq 2$

Rule 1.2.1.4.10.2.1.1.3: If $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,m\in\mathbb{Z}\,\wedge\,m\geq 2$**, then**

$$\int \frac{(d+e\,x)^m}{\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx\,\longrightarrow$$

$$\frac{2\,e^2\,(d+e\,x)^{m-2}\,\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}{c\,g\,(2\,m-1)}-$$
$$\frac{1}{c\,g\,(2\,m-1)}\int \frac{(d+e\,x)^{m-3}}{\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,\cdot$$
$$\left(b\,d\,e^2\,f+a\,e^2\,(d\,g+2\,e\,f\,(m-2))-c\,d^3\,g\,(2\,m-1)+\right.$$
$$\left.e\,(e\,(2\,b\,d\,g+e\,(b\,f+a\,g)\,(2\,m-3))+c\,d\,(2\,e\,f-3\,d\,g\,(2\,m-1)))\,x+2\,e^2\,(c\,e\,f-3\,c\,d\,g+b\,e\,g)\,(m-1)\,x^2\right)dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_/(Sqrt[f_.+g_.*x_]*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  2*e^2*(d+e*x)^(m-2)*Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2]/(c*g*(2*m-1)) -
  1/(c*g*(2*m-1))*Int[(d+e*x)^(m-3)/(Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2])*
    Simp[b*d*e^2*f+a*e^2*(d*g+2*e*f*(m-2))-c*d^3*g*(2*m-1)+
      e*(e*(2*b*d*g+e*(b*f+a*g)*(2*m-3))+c*d*(2*e*f-3*d*g*(2*m-1)))*x+
      2*e^2*(c*e*f-3*c*d*g+b*e*g)*(m-1)*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[2*m] && GeQ[m,2]
```

```
Int[(d_.+e_.*x_)^m_/(Sqrt[f_.+g_.*x_]*Sqrt[a_+c_.*x_^2]),x_Symbol] :=
  2*e^2*(d+e*x)^(m-2)*Sqrt[f+g*x]*Sqrt[a+c*x^2]/(c*g*(2*m-1)) -
  1/(c*g*(2*m-1))*Int[(d+e*x)^(m-3)/(Sqrt[f+g*x]*Sqrt[a+c*x^2])*
    Simp[a*e^2*(d*g+2*e*f*(m-2))-c*d^3*g*(2*m-1)+e*(e*(a*e*g*(2*m-3))+c*d*(2*e*f-3*d*g*(2*m-1)))*x+2*e^2*(c*e*f-3*c*d*g)*(m-1)*x^2,
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && IntegerQ[2*m] && GeQ[m,2]
```

2. $\int \frac{(d+e\,x)^m}{\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,m\in\mathbb{Z}\,\wedge\,m<0$

1. $\int \frac{1}{(d+e\,x)\,\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0$

1. $\int \frac{1}{(d+e\,x)\,\sqrt{f+g\,x}\,\sqrt{a+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\land\,c\,d^2+a\,e^2\neq 0$

1: $\int \frac{1}{(d+e\,x)\,\sqrt{f+g\,x}\,\sqrt{a+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\land\,c\,d^2+a\,e^2\neq 0\,\land\,a>0$

Derivation: Algebraic expansion

- **Basis: If** $a>0$**, let** $q\to\sqrt{-\frac{c}{a}}$**, then** $\sqrt{a+c\,x^2}==\sqrt{a}\,\sqrt{1-q\,x}\,\sqrt{1+q\,x}$
- **Rule 1.2.1.4.10.2.1.2.1.1.1: If** $e\,f-d\,g\neq 0\,\land\,c\,d^2+a\,e^2\neq 0\,\land\,a>0$**, let** $q\to\sqrt{-\frac{c}{a}}$**, then**

$$\int \frac{1}{(d+e\,x)\,\sqrt{f+g\,x}\,\sqrt{a+c\,x^2}}\,dx\;\longrightarrow\;\frac{1}{\sqrt{a}}\int \frac{1}{(d+e\,x)\,\sqrt{f+g\,x}\,\sqrt{1-q\,x}\,\sqrt{1+q\,x}}\,dx$$

Program code:

```
Int[1/((d_.+e_.*x_)*Sqrt[f_.+g_.*x_]*Sqrt[a_+c_.*x_^2]),x_Symbol] :=
  With[{q=Rt[-c/a,2]},
  1/Sqrt[a]*Int[1/((d+e*x)*Sqrt[f+g*x]*Sqrt[1-q*x]*Sqrt[1+q*x]),x]] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && GtQ[a,0]
```

2: $\int \frac{1}{(d+e\,x)\,\sqrt{f+g\,x}\,\sqrt{a+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,c\,d^2+a\,e^2\neq 0\,\wedge\,a\ngtr 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{1+\frac{c\,x^2}{a}}}{\sqrt{a+c\,x^2}} == 0$
- **Basis: Let** $q\to\sqrt{-\frac{c}{a}}$**, then** $\sqrt{1+\frac{c\,x^2}{a}} == \sqrt{1-q\,x}\,\sqrt{1+q\,x}$
- **Rule 1.2.1.4.10.2.1.2.1.1.2: If** $e\,f-d\,g\neq 0\,\wedge\,c\,d^2+a\,e^2\neq 0\,\wedge\,a\ngtr 0$**, let** $q\to\sqrt{-\frac{c}{a}}$**, then**

$$\int \frac{1}{(d+e\,x)\,\sqrt{f+g\,x}\,\sqrt{a+c\,x^2}}\,dx\,\longrightarrow\,\frac{\sqrt{1+\frac{c\,x^2}{a}}}{\sqrt{a+c\,x^2}}\int \frac{1}{(d+e\,x)\,\sqrt{f+g\,x}\,\sqrt{1-q\,x}\,\sqrt{1+q\,x}}\,dx$$

Program code:

```
Int[1/((d_.+e_.*x_)*Sqrt[f_.+g_.*x_]*Sqrt[a_+c_.*x_^2]),x_Symbol] :=
  With[{q=Rt[-c/a,2]},
  Sqrt[1+c*x^2/a]/Sqrt[a+c*x^2]*Int[1/((d+e*x)*Sqrt[f+g*x]*Sqrt[1-q*x]*Sqrt[1+q*x]),x]] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && Not[GtQ[a,0]]
```

2: $\int \frac{1}{(d+e\,x)\,\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0$

Derivation: Piecewise constant extraction

- **Basis: Let** $q\to\sqrt{b^2-4\,a\,c}$ **, then** $\partial_x\frac{\sqrt{b-q+2\,c\,x}\,\sqrt{b+q+2\,c\,x}}{\sqrt{a+b\,x+c\,x^2}}==0$
- **Rule 1.2.1.4.10.2.1.2.1.2: If** $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0$**, let** $q\to\sqrt{b^2-4\,a\,c}$ **, then**

$$\int \frac{1}{(d+e\,x)\,\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx \longrightarrow \frac{\sqrt{b-q+2\,c\,x}\,\sqrt{b+q+2\,c\,x}}{\sqrt{a+b\,x+c\,x^2}}\int \frac{1}{(d+e\,x)\,\sqrt{f+g\,x}\,\sqrt{b-q+2\,c\,x}\,\sqrt{b+q+2\,c\,x}}\,dx$$

Program code:

```
Int[1/((d_.+e_.*x_)*Sqrt[f_.+g_.*x_]*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  Sqrt[b-q+2*c*x]*Sqrt[b+q+2*c*x]/Sqrt[a+b*x+c*x^2]*Int[1/((d+e*x)*Sqrt[f+g*x]*Sqrt[b-q+2*c*x]*Sqrt[b+q+2*c*x]),x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0]
```

2: $\int \frac{1}{\sqrt{d+e\,x}\,\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x\frac{(d+e\,x)\sqrt{\frac{(e\,f-d\,g)^2\,(a+b\,x+c\,x^2)}{(c\,f^2-b\,f\,g+a\,g^2)\,(d+e\,x)^2}}}{\sqrt{a+b\,x+c\,x^2}}==0$
- **Basis:** $\frac{1}{(d+e\,x)^{3/2}\sqrt{f+g\,x}\sqrt{\frac{(e\,f-d\,g)^2\,(a+b\,x+c\,x^2)}{(c\,f^2-b\,f\,g+a\,g^2)\,(d+e\,x)^2}}}==-\frac{2}{e\,f-d\,g}\,\text{Subst}\left[\frac{1}{\sqrt{1-\frac{(2\,c\,d\,f-b\,e\,f-b\,d\,g+2\,a\,e\,g)\,x^2}{c\,f^2-b\,f\,g+a\,g^2}+\frac{(c\,d^2-b\,d\,e+a\,e^2)\,x^4}{c\,f^2-b\,f\,g+a\,g^2}}},\,x,\,\frac{\sqrt{f+g\,x}}{\sqrt{d+e\,x}}\right]\partial_x\frac{\sqrt{f+g\,x}}{\sqrt{d+e\,x}}$
- **Rule 1.2.1.4.10.2.1.2.2: If** $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0$**, then**

$$\int \frac{1}{\sqrt{d+e\,x}\,\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx \longrightarrow \frac{(d+e\,x)\sqrt{\frac{(e\,f-d\,g)^2\,(a+b\,x+c\,x^2)}{(c\,f^2-b\,f\,g+a\,g^2)\,(d+e\,x)^2}}}{\sqrt{a+b\,x+c\,x^2}}\int \frac{1}{(d+e\,x)^{3/2}\,\sqrt{f+g\,x}\,\sqrt{\frac{(e\,f-d\,g)^2\,(a+b\,x+c\,x^2)}{(c\,f^2-b\,f\,g+a\,g^2)\,(d+e\,x)^2}}}\,dx$$

$$\to -\frac{2\,(d+e\,x)\,\sqrt{\frac{(e\,f-d\,g)^2\,\left(a+b\,x+c\,x^2\right)}{\left(c\,f^2-b\,f\,g+a\,g^2\right)\,(d+e\,x)^2}}}{(e\,f-d\,g)\,\sqrt{a+b\,x+c\,x^2}}\,\text{Subst}\!\left[\int\frac{1}{\sqrt{1-\frac{(2\,c\,d\,f-b\,e\,f-b\,d\,g+2\,a\,e\,g)\,x^2}{c\,f^2-b\,f\,g+a\,g^2}+\frac{\left(c\,d^2-b\,d\,e+a\,e^2\right)\,x^4}{c\,f^2-b\,f\,g+a\,g^2}}}\,dx,\,x,\,\frac{\sqrt{f+g\,x}}{\sqrt{d+e\,x}}\right]$$

– **Program code:**

```
Int[1/(Sqrt[d_.+e_.*x_]*Sqrt[f_.+g_.*x_]*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  -2*(d+e*x)*Sqrt[(e*f-d*g)^2*(a+b*x+c*x^2)/((c*f^2-b*f*g+a*g^2)*(d+e*x)^2)]/((e*f-d*g)*Sqrt[a+b*x+c*x^2])*
  Subst[
    Int[1/Sqrt[1-(2*c*d*f-b*e*f-b*d*g+2*a*e*g)*x^2/(c*f^2-b*f*g+a*g^2)+(c*d^2-b*d*e+a*e^2)*x^4/(c*f^2-b*f*g+a*g^2)],x],
    x,
    Sqrt[f+g*x]/Sqrt[d+e*x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[1/(Sqrt[d_.+e_.*x_]*Sqrt[f_.+g_.*x_]*Sqrt[a_+c_.*x_^2]),x_Symbol] :=
  -2*(d+e*x)*Sqrt[(e*f-d*g)^2*(a+c*x^2)/((c*f^2+a*g^2)*(d+e*x)^2)]/((e*f-d*g)*Sqrt[a+c*x^2])*
  Subst[
    Int[1/Sqrt[1-(2*c*d*f+2*a*e*g)*x^2/(c*f^2+a*g^2)+(c*d^2+a*e^2)*x^4/(c*f^2+a*g^2)],x],x,Sqrt[f+g*x]/Sqrt[d+e*x]] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0]
```

3: $\int \frac{1}{(d+e\,x)^{3/2}\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0$

Derivation: Algebraic expansion

- **Basis:** $\frac{1}{(d+e\,x)^{3/2}\sqrt{f+g\,x}} == -\frac{g}{(e\,f-d\,g)\sqrt{d+e\,x}\,\sqrt{f+g\,x}}+\frac{e\sqrt{f+g\,x}}{(e\,f-d\,g)\,(d+e\,x)^{3/2}}$

Rule 1.2.1.4.10.2.1.2.3: If when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0$**, then**

$$\int \frac{1}{(d+e\,x)^{3/2}\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx\,\longrightarrow$$

$$-\frac{g}{e\,f-d\,g}\int \frac{1}{\sqrt{d+e\,x}\,\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx+\frac{e}{e\,f-d\,g}\int \frac{\sqrt{f+g\,x}}{(d+e\,x)^{3/2}\sqrt{a+b\,x+c\,x^2}}\,dx$$

Program code:

```
Int[1/((d_.+e_.*x_)^(3/2)*Sqrt[f_.+g_.*x_]*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  -g/(e*f-d*g)*Int[1/(Sqrt[d+e*x]*Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2]),x] +
  e/(e*f-d*g)*Int[Sqrt[f+g*x]/((d+e*x)^(3/2)*Sqrt[a+b*x+c*x^2]),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[1/((d_.+e_.*x_)^(3/2)*Sqrt[f_.+g_.*x_]*Sqrt[a_+c_.*x_^2]),x_Symbol] :=
  -g/(e*f-d*g)*Int[1/(Sqrt[d+e*x]*Sqrt[f+g*x]*Sqrt[a+c*x^2]),x] +
  e/(e*f-d*g)*Int[Sqrt[f+g*x]/((d+e*x)^(3/2)*Sqrt[a+c*x^2]),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0]
```

4: $\int \frac{(d+e\,x)^m}{\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,m\in\mathbb{Z}\,\wedge\,m\leq -2$

Rule 1.2.1.4.10.2.1.2.4: If $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,m\in\mathbb{Z}\,\wedge\,m\leq -2$**, then**

$$\int \frac{(d+e\,x)^m}{\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx\,\longrightarrow$$

$$\frac{e^2 (d+e x)^{m+1} \sqrt{f+g x} \sqrt{a+b x+c x^2}}{(m+1) (e f-d g) \left(c d^2-b d e+a e^2\right)}+$$
$$\frac{1}{2 (m+1) (e f-d g) \left(c d^2-b d e+a e^2\right)} \int \frac{(d+e x)^{m+1}}{\sqrt{f+g x} \sqrt{a+b x+c x^2}} \cdot$$
$$\left(2 d (c e f-c d g+b e g) (m+1)-e^2 (b f+a g) (2 m+3)+2 e (c d g (m+1)-e (c f+b g) (m+2)) x-c e^2 g (2 m+5) x^2\right) dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_/(Sqrt[f_.+g_.*x_]*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  e^2*(d+e*x)^(m+1)*Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2]/((m+1)*(e*f-d*g)*(c*d^2-b*d*e+a*e^2)) +
  1/(2*(m+1)*(e*f-d*g)*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x)^(m+1)/(Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2])*
    Simp[2*d*(c*e*f-c*d*g+b*e*g)*(m+1)-e^2*(b*f+a*g)*(2*m+3)+2*e*(c*d*g*(m+1)-e*(c*f+b*g)*(m+2))*x-c*e^2*g*(2*m+5)*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[2*m] && LeQ[m,-2]
```

```
Int[(d_.+e_.*x_)^m_/(Sqrt[f_.+g_.*x_]*Sqrt[a_+c_.*x_^2]),x_Symbol] :=
  e^2*(d+e*x)^(m+1)*Sqrt[f+g*x]*Sqrt[a+c*x^2]/((m+1)*(e*f-d*g)*(c*d^2+a*e^2)) +
  1/(2*(m+1)*(e*f-d*g)*(c*d^2+a*e^2))*Int[(d+e*x)^(m+1)/(Sqrt[f+g*x]*Sqrt[a+c*x^2])*
    Simp[2*d*(c*e*f-c*d*g)*(m+1)-a*e^2*g*(2*m+3)+2*e*(c*d*g*(m+1)-c*e*f*(m+2))*x-c*e^2*g*(2*m+5)*x^2,x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && IntegerQ[2*m] && LeQ[m,-2]
```

2. $\int \frac{(d+e x)^m \sqrt{f+g x}}{\sqrt{a+b x+c x^2}} dx$ when $e f-d g \neq 0 \land b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 m \in \mathbb{Z}$

   1. $\int \frac{(d+e x)^m \sqrt{f+g x}}{\sqrt{a+b x+c x^2}} dx$ when $e f-d g \neq 0 \land b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 m \in \mathbb{Z} \land m>0$

      x: $\int \frac{\sqrt{d+e x} \sqrt{f+g x}}{\sqrt{a+b x+c x^2}} dx$ when $e f-d g \neq 0 \land b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0$

Derivation: Algebraic expansion

–

Rule 1.2.1.4.10.2.2.1.x: If $e f-d g \neq 0 \land b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0$**, then**

$$\int \frac{\sqrt{d+e x} \sqrt{f+g x}}{\sqrt{a+b x+c x^2}} dx \longrightarrow$$

$$\frac{\sqrt{a+b x}\ \sqrt{c+d x}\ \sqrt{g+h x}}{h \sqrt{e+f x}}+\frac{(d e-c f)\ (b f g+b e h-2 a f h)}{2 f^2 h} \int \frac{1}{\sqrt{a+b x}\ \sqrt{c+d x}\ \sqrt{e+f x}\ \sqrt{g+h x}} \, dx+$$

$$\frac{(a d f h-b\ (d f g+d e h-c f h))}{2 f^2 h} \int \frac{\sqrt{e+f x}}{\sqrt{a+b x}\ \sqrt{c+d x}\ \sqrt{g+h x}} \, dx-\frac{(d e-c f)\ (f g-e h)}{2 f h} \int \frac{\sqrt{a+b x}}{\sqrt{c+d x}\ (e+f x)^{3/2} \sqrt{g+h x}} \, dx$$

– **Program code:**

```
(* Int[Sqrt[d_.+e_.*x_]*Sqrt[f_.+g_.*x_]/Sqrt[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  0 /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] *)
```

2: $\int \frac{(d+e x)^m \sqrt{f+g x}}{\sqrt{a+b x+c x^2}} \, dx$ when $e f-d g \neq 0 \land b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 m \in \mathbb{Z} \land m>1$

Rule 1.2.1.4.10.2.2.1.2: If $e f-d g \neq 0 \land b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land 2 m \in \mathbb{Z} \land m>1$**, then**

$$\int \frac{(d+e x)^m \sqrt{f+g x}}{\sqrt{a+b x+c x^2}} \, dx \longrightarrow$$

$$\frac{2 e\ (d+e x)^{m-1} \sqrt{f+g x}\ \sqrt{a+b x+c x^2}}{c\ (2 m+1)}-\frac{1}{c\ (2 m+1)} \int \frac{(d+e x)^{m-2}}{\sqrt{f+g x}\ \sqrt{a+b x+c x^2}} \cdot$$

$$\Big(e\ (b d f+a\ (d g+2 e f\ (m-1)))-c d^2 f\ (2 m+1)+$$
$$\left(a e^2 g\ (2 m-1)-c d\ (4 e f m+d g\ (2 m+1))+b e\ (2 d g+e f\ (2 m-1))\right) x+e\ (2 b e g m-c\ (e f+d g\ (4 m-1)))\ x^2\Big) \, dx$$

– **Program code:**

```
Int[(d_.+e_.*x_)^m_*Sqrt[f_.+g_.*x_]/Sqrt[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  2*e*(d+e*x)^(m-1)*Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2]/(c*(2*m+1)) -
  1/(c*(2*m+1))*Int[(d+e*x)^(m-2)/(Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2])*
    Simp[e*(b*d*f+a*(d*g+2*e*f*(m-1)))-c*d^2*f*(2*m+1)+
      (a*e^2*g*(2*m-1)-c*d*(4*e*f*m+d*g*(2*m+1))+b*e*(2*d*g+e*f*(2*m-1)))*x+
      e*(2*b*e*g*m-c*(e*f+d*g*(4*m-1)))*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[2*m] && GtQ[m,1]
```

```
Int[(d_.+e_.*x_)^m_*Sqrt[f_.+g_.*x_]/Sqrt[a_+c_.*x_^2],x_Symbol] :=
  2*e*(d+e*x)^(m-1)*Sqrt[f+g*x]*Sqrt[a+c*x^2]/(c*(2*m+1)) -
  1/(c*(2*m+1))*Int[(d+e*x)^(m-2)/(Sqrt[f+g*x]*Sqrt[a+c*x^2])*
    Simp[a*e*(d*g+2*e*f*(m-1))-c*d^2*f*(2*m+1)+(a*e^2*g*(2*m-1)-c*d*(4*e*f*m+d*g*(2*m+1)))*x-c*e*(e*f+d*g*(4*m-1))*x^2,x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && IntegerQ[2*m] && GtQ[m,1]
```

2. $\int \frac{(d+e\,x)^m \sqrt{f+g\,x}}{\sqrt{a+b\,x+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,2\,m\in\mathbb{Z}\,\wedge\,m<0$

1: $\int \frac{\sqrt{f+g\,x}}{(d+e\,x)\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0$

Derivation: Algebraic expansion

- **Basis:** $\frac{\sqrt{f+g\,x}}{d+e\,x} == \frac{g}{e\,\sqrt{f+g\,x}}+\frac{e\,f-d\,g}{e\,(d+e\,x)\,\sqrt{f+g\,x}}$

Rule 1.2.1.4.10.2.2.2.1: If $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0$**, then**

$$\int \frac{\sqrt{f+g\,x}}{(d+e\,x)\,\sqrt{a+b\,x+c\,x^2}}\,dx \to \frac{g}{e}\int \frac{1}{\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx+\frac{(e\,f-d\,g)}{e}\int \frac{1}{(d+e\,x)\,\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$$

Program code:

```
Int[Sqrt[f_.+g_.*x_]/((d_.+e_.*x_)*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  g/e*Int[1/(Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2]),x] +
  (e*f-d*g)/e*Int[1/((d+e*x)*Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2]),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[Sqrt[f_.+g_.*x_]/((d_.+e_.*x_)*Sqrt[a_+c_.*x_^2]),x_Symbol] :=
  g/e*Int[1/(Sqrt[f+g*x]*Sqrt[a+c*x^2]),x] +
  (e*f-d*g)/e*Int[1/((d+e*x)*Sqrt[f+g*x]*Sqrt[a+c*x^2]),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0]
```

x: $\int \frac{\sqrt{f+g\,x}}{(d+e\,x)^{3/2}\,\sqrt{a+b\,x+c\,x^2}}\,dx$

3: $\int \frac{(d+e\,x)^m \sqrt{f+g\,x}}{\sqrt{a+b\,x+c\,x^2}}\,dx$ when $e\,f-d\,g\neq 0\,\land\,b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,2\,m\in\mathbb{Z}\,\land\,m\leq -2$

Rule 1.2.1.4.10.2.2.2.3: If $e\,f-d\,g\neq 0\,\land\,b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,2\,m\in\mathbb{Z}\,\land\,m\leq -2$**, then**

$$\int \frac{(d+e\,x)^m \sqrt{f+g\,x}}{\sqrt{a+b\,x+c\,x^2}}\,dx \longrightarrow$$

$$\frac{e\,(d+e\,x)^{m+1}\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}{(m+1)\left(c\,d^2-b\,d\,e+a\,e^2\right)}+\frac{1}{2\,(m+1)\left(c\,d^2-b\,d\,e+a\,e^2\right)}\int \frac{(d+e\,x)^{m+1}}{\sqrt{f+g\,x}\,\sqrt{a+b\,x+c\,x^2}}\cdot$$
$$\left(2\,c\,d\,f\,(m+1)-e\,(a\,g+b\,f\,(2\,m+3))-2\,(b\,e\,g\,(2+m)-c\,(d\,g\,(m+1)-e\,f\,(m+2)))\,x-c\,e\,g\,(2\,m+5)\,x^2\right)dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*Sqrt[f_.+g_.*x_]/Sqrt[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*(d+e*x)^(m+1)*Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2]/((m+1)*(c*d^2-b*d*e+a*e^2)) +
  1/(2*(m+1)*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x)^(m+1)/(Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2])*
    Simp[2*c*d*f*(m+1)-e*(a*g+b*f*(2*m+3))-2*(b*e*g*(2+m)-c*(d*g*(m+1)-e*f*(m+2)))*x-c*e*g*(2*m+5)*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[2*m] && LeQ[m,-2]
```

```
Int[(d_.+e_.*x_)^m_*Sqrt[f_.+g_.*x_]/Sqrt[a_+c_.*x_^2],x_Symbol] :=
  e*(d+e*x)^(m+1)*Sqrt[f+g*x]*Sqrt[a+c*x^2]/((m+1)*(c*d^2+a*e^2)) +
  1/(2*(m+1)*(c*d^2+a*e^2))*Int[(d+e*x)^(m+1)/(Sqrt[f+g*x]*Sqrt[a+c*x^2])*
    Simp[2*c*d*f*(m+1)-e*(a*g)+2*c*(d*g*(m+1)-e*f*(m+2))*x-c*e*g*(2*m+5)*x^2,x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && IntegerQ[2*m] && LeQ[m,-2]
```

11. $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,p\in\mathbb{Z}^+$

1: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,p\in\mathbb{Z}^+\,\wedge\,m\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.1.4.11.1: If $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,p\in\mathbb{Z}^+\,\wedge\,m\in\mathbb{Z}^+$**, then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\int \text{ExpandIntegrand}\left[(d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p,\,x\right]dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)^n*(a+b*x+c*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IGtQ[p,0] &&
  (IGtQ[m,0] || EqQ[m,-2] && EqQ[p,1] && EqQ[2*c*d-b*e,0])
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)^n*(a+c*x^2)^p,x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && IGtQ[p,0] &&
  (IGtQ[m,0] || EqQ[m,-2] && EqQ[p,1] && EqQ[d,0])
```

2: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,p\in\mathbb{Z}^+\wedge\,m<-1$

Derivation: Algebraic expansion and linear recurrence 3

– **Basis: Let** $Q[x]\to\text{PolynomialQuotient}\left[\left(a+b\,x+c\,x^2\right)^p,\,d+e\,x,\,x\right]$ **and** $R\to\text{PolynomialRemainder}\left[\left(a+b\,x+c\,x^2\right)^p,\,d+e\,x,\,x\right]$,
then $\left(a+b\,x+c\,x^2\right)^p == Q[x]\,(d+e\,x)+R$

Note: If $m\in\mathbb{Z}^-$**, incrementing** m **rather than** n **produces simpler antiderivatives.**

– **Rule 1.2.1.4.11.2: If** $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,p\in\mathbb{Z}^+\wedge\,m<-1$,
let $Q[x]\to\text{PolynomialQuotient}\left[\left(a+b\,x+c\,x^2\right)^p,\,d+e\,x,\,x\right]$ **and** $R\to\text{PolynomialRemainder}\left[\left(a+b\,x+c\,x^2\right)^p,\,d+e\,x,\,x\right]$**, then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$

$$\int Q[x]\,(d+e\,x)^{m+1}\,(f+g\,x)^n\,dx+R\int (d+e\,x)^m\,(f+g\,x)^n\,dx\,\longrightarrow$$

$$\frac{R\,(d+e\,x)^{m+1}\,(f+g\,x)^{n+1}}{(m+1)\,(e\,f-d\,g)}+\frac{1}{(m+1)\,(e\,f-d\,g)}\int(d+e\,x)^{m+1}\,(f+g\,x)^n\,((m+1)\,(e\,f-d\,g)\,Q[x]-g\,R\,(m+n+2))\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  With[{Qx=PolynomialQuotient[(a+b*x+c*x^2)^p,d+e*x,x],R=PolynomialRemainder[(a+b*x+c*x^2)^p,d+e*x,x]},
  R*(d+e*x)^(m+1)*(f+g*x)^(n+1)/((m+1)*(e*f-d*g)) +
  1/((m+1)*(e*f-d*g))*Int[(d+e*x)^(m+1)*(f+g*x)^n*ExpandToSum[(m+1)*(e*f-d*g)*Qx-g*R*(m+n+2),x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IGtQ[p,0] && LtQ[m,-1]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  With[{Qx=PolynomialQuotient[(a+c*x^2)^p,d+e*x,x],R=PolynomialRemainder[(a+c*x^2)^p,d+e*x,x]},
  R*(d+e*x)^(m+1)*(f+g*x)^(n+1)/((m+1)*(e*f-d*g)) +
  1/((m+1)*(e*f-d*g))*Int[(d+e*x)^(m+1)*(f+g*x)^n*ExpandToSum[(m+1)*(e*f-d*g)*Qx-g*R*(m+n+2),x],x]] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && IGtQ[p,0] && LtQ[m,-1]
```

3: $\int(d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq0\,\wedge\,b^2-4\,a\,c\neq0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq0\,\wedge\,p\in\mathbb{Z}^+\wedge\,m+n+2\,p+1\neq0$

Derivation: Algebraic expansion and linear recurrence 2

Rule 1.2.1.4.11.3: If $e\,f-d\,g\neq0\,\wedge\,b^2-4\,a\,c\neq0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq0\,\wedge\,p\in\mathbb{Z}^+\wedge\,m+n+2\,p+1\neq0$**, then**

$$\int(d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$

$$\frac{1}{e^{2p}}\int\left(e^{2p}\,\left(a+b\,x+c\,x^2\right)^p-c^p\,(d+e\,x)^{2p}\right)\,(d+e\,x)^m\,(f+g\,x)^n\,dx+\frac{c^p}{e^{2p}}\int(d+e\,x)^{m+2p}\,(f+g\,x)^n\,dx\,\longrightarrow$$

$$\frac{c^p\,(d+e\,x)^{m+2p}\,(f+g\,x)^{n+1}}{g\,e^{2p}\,(m+n+2\,p+1)}+\frac{1}{g\,e^{2p}\,(m+n+2\,p+1)}\int(d+e\,x)^m\,(f+g\,x)^n\cdot$$
$$\left(g\,(m+n+2\,p+1)\,\left(e^{2p}\,\left(a+b\,x+c\,x^2\right)^p-c^p\,(d+e\,x)^{2p}\right)-c^p\,(e\,f-d\,g)\,(m+2\,p)\,(d+e\,x)^{2p-1}\right)\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  c^p*(d+e*x)^(m+2*p)*(f+g*x)^(n+1)/(g*e^(2*p)*(m+n+2*p+1)) +
  1/(g*e^(2*p)*(m+n+2*p+1))*Int[(d+e*x)^m*(f+g*x)^n*
    ExpandToSum[g*(m+n+2*p+1)*(e^(2*p)*(a+b*x+c*x^2)^p-c^p*(d+e*x)^(2*p))-c^p*(e*f-d*g)*(m+2*p)*(d+e*x)^(2*p-1),x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IGtQ[p,0] && NeQ[m+n+2*p+1,0] &&
  (IntegerQ[n] || Not[IntegerQ[m]])
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  c^p*(d+e*x)^(m+2*p)*(f+g*x)^(n+1)/(g*e^(2*p)*(m+n+2*p+1)) +
  1/(g*e^(2*p)*(m+n+2*p+1))*Int[(d+e*x)^m*(f+g*x)^n*
    ExpandToSum[g*(m+n+2*p+1)*(e^(2*p)*(a+c*x^2)^p-c^p*(d+e*x)^(2*p))-c^p*(e*f-d*g)*(m+2*p)*(d+e*x)^(2*p-1),x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && IGtQ[p,0] && NeQ[m+n+2*p+1,0] &&
  (IntegerQ[n] || Not[IntegerQ[m]])
```

12. $\int \frac{(f+g x)^n \left(a+b x+c x^2\right)^p}{d+e x} \, dx$ when $e f-d g\neq 0 \land b^2-4 a c\neq 0 \land c d^2-b d e+a e^2\neq 0 \land n\notin \mathbb{Z} \land p\notin \mathbb{Z}$

1: $\int \frac{(f+g x)^n \left(a+b x+c x^2\right)^p}{d+e x} \, dx$ when $e f-d g\neq 0 \land b^2-4 a c\neq 0 \land c d^2-b d e+a e^2\neq 0 \land n\notin \mathbb{Z} \land p\notin \mathbb{Z} \land p>0 \land n<-1$

Reference: Algebraic expansion

- Basis: $\frac{a+b x+c x^2}{d+e x} == \frac{\left(c d^2-b d e+a e^2\right) (f+g x)}{e (e f-d g) (d+e x)} - \frac{c d f-b e f+a e g-c (e f-d g) x}{e (e f-d g)}$

Rule 1.2.1.4.12.1: If $e f-d g\neq 0 \land b^2-4 a c\neq 0 \land c d^2-b d e+a e^2\neq 0 \land n\notin \mathbb{Z} \land p\notin \mathbb{Z} \land p>0 \land n<-1$**, then**

$$\int \frac{(f+g x)^n \left(a+b x+c x^2\right)^p}{d+e x} \, dx \longrightarrow$$

$$\frac{c d^2-b d e+a e^2}{e (e f-d g)} \int \frac{(f+g x)^{n+1} \left(a+b x+c x^2\right)^{p-1}}{d+e x} \, dx - \frac{1}{e (e f-d g)} \int (f+g x)^n (c d f-b e f+a e g-c (e f-d g) x) \left(a+b x+c x^2\right)^{p-1} dx$$

Program code:

```
Int[(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_/(d_.+e_.*x_),x_Symbol] :=
  (c*d^2-b*d*e+a*e^2)/(e*(e*f-d*g))*Int[(f+g*x)^(n+1)*(a+b*x+c*x^2)^(p-1)/(d+e*x),x] -
  1/(e*(e*f-d*g))*Int[(f+g*x)^n*(c*d*f-b*e*f+a*e*g-c*(e*f-d*g)*x)*(a+b*x+c*x^2)^(p-1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IntegerQ[n]] && Not[IntegerQ[p]] && GtQ[p,0] && LtQ[n,-1]
```

```
Int[(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_/(d_.+e_.*x_),x_Symbol] :=
  (c*d^2+a*e^2)/(e*(e*f-d*g))*Int[(f+g*x)^(n+1)*(a+c*x^2)^(p-1)/(d+e*x),x] -
  1/(e*(e*f-d*g))*Int[(f+g*x)^n*(c*d*f+a*e*g-c*(e*f-d*g)*x)*(a+c*x^2)^(p-1),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] &&
  Not[IntegerQ[n]] && Not[IntegerQ[p]] && GtQ[p,0] && LtQ[n,-1]
```

2: $\int \frac{(f+g x)^n \left(a+b x+c x^2\right)^p}{d+e x} \, dx$ when $e f-d g \neq 0 \wedge b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge n \notin \mathbb{Z} \wedge p \notin \mathbb{Z} \wedge p<-1 \wedge n>0$

Reference: Algebraic expansion

- **Basis:** $\frac{f+g x}{d+e x} = \frac{e\,(e f-d g)\,(a+b x+c x^2)}{(c d^2-b d e+a e^2)\,(d+e x)} + \frac{c d f-b e f+a e g-c\,(e f-d g)\,x}{c d^2-b d e+a e^2}$

Rule 1.2.1.4.12.2: If $e f-d g \neq 0 \wedge b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge n \notin \mathbb{Z} \wedge p \notin \mathbb{Z} \wedge p<-1 \wedge n>0$**, then**

$$\int \frac{(f+g x)^n \left(a+b x+c x^2\right)^p}{d+e x} \, dx \longrightarrow$$

$$\frac{e\,(e f-d g)}{c d^2-b d e+a e^2} \int \frac{(f+g x)^{n-1} \left(a+b x+c x^2\right)^{p+1}}{d+e x} \, dx + \frac{1}{c d^2-b d e+a e^2} \int (f+g x)^{n-1} \left(c d f-b e f+a e g-c\,(e f-d g)\,x\right) \left(a+b x+c x^2\right)^p \, dx$$

Program code:

```
Int[(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_/(d_.+e_.*x_),x_Symbol] :=
  e*(e*f-d*g)/(c*d^2-b*d*e+a*e^2)*Int[(f+g*x)^(n-1)*(a+b*x+c*x^2)^(p+1)/(d+e*x),x] +
  1/(c*d^2-b*d*e+a*e^2)*Int[(f+g*x)^(n-1)*(c*d*f-b*e*f+a*e*g-c*(e*f-d*g)*x)*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IntegerQ[n]] && Not[IntegerQ[p]] && LtQ[p,-1] && GtQ[n,0]
```

```
Int[(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_/(d_.+e_.*x_),x_Symbol] :=
  e*(e*f-d*g)/(c*d^2+a*e^2)*Int[(f+g*x)^(n-1)*(a+c*x^2)^(p+1)/(d+e*x),x] +
  1/(c*d^2+a*e^2)*Int[(f+g*x)^(n-1)*(c*d*f+a*e*g-c*(e*f-d*g)*x)*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] &&
  Not[IntegerQ[n]] && Not[IntegerQ[p]] && LtQ[p,-1] && GtQ[n,0]
```

3: $\int \frac{(f+g x)^n}{(d+e x)\sqrt{a+b x+c x^2}}\,dx$ when $e f-d g\neq 0\,\bigwedge\, b^2-4 a c\neq 0\,\bigwedge\, c d^2-b d e+a e^2\neq 0\,\bigwedge\, n+\frac{1}{2}\in\mathbb{Z}$

Reference: Algebraic expansion

- **Rule 1.2.1.4.12.3: If** $e f-d g\neq 0\,\bigwedge\, b^2-4 a c\neq 0\,\bigwedge\, c d^2-b d e+a e^2\neq 0\,\bigwedge\, n+\frac{1}{2}\in\mathbb{Z}$**, then**

$$\int \frac{(f+g x)^n}{(d+e x)\sqrt{a+b x+c x^2}}\,dx \to \int \frac{1}{\sqrt{f+g x}\,\sqrt{a+b x+c x^2}}\,\text{ExpandIntegrand}\left[\frac{(f+g x)^{n+\frac{1}{2}}}{d+e x},x\right]dx$$

Program code:

```
Int[(f_.+g_.*x_)^n_/((d_.+e_.*x_)*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  Int[ExpandIntegrand[1/(Sqrt[f+g*x]*Sqrt[a+b*x+c*x^2]),(f+g*x)^(n+1/2)/(d+e*x),x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[n+1/2]
```

```
Int[(f_.+g_.*x_)^n_/((d_.+e_.*x_)*Sqrt[a_+c_.*x_^2]),x_Symbol] :=
  Int[ExpandIntegrand[1/(Sqrt[f+g*x]*Sqrt[a+c*x^2]),(f+g*x)^(n+1/2)/(d+e*x),x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && IntegerQ[n+1/2]
```

13: $\int \frac{(g x)^n\,(a+c x^2)^p}{d+e x}\,dx$ when $c d^2+a e^2\neq 0\,\wedge\, p\notin\mathbb{Z}\,\wedge\,\neg\,(n\in\mathbb{Z}\,\wedge\, 2 p\in\mathbb{Z})$

Derivation: Algebraic expansion

- **Basis:** $\frac{1}{d+e x}==\frac{d}{d^2-e^2 x^2}-\frac{e x}{d^2-e^2 x^2}$

- **Note: Resulting integrands are of the form** $\frac{x^m\,(a+b x^2)^p}{c+d x^2}$ **which are integrable in terms of the Appell hypergeometric function .**

Rule 1.2.1.4.13: If $c d^2+a e^2\neq 0\,\wedge\, p\notin\mathbb{Z}\,\wedge\,\neg\,(n\in\mathbb{Z}\,\wedge\, 2 p\in\mathbb{Z})$**, then**

$$\int \frac{(g x)^n\,(a+c x^2)^p}{d+e x}\,dx \to \frac{d\,(g x)^n}{x^n}\int \frac{x^n\,(a+c x^2)^p}{d^2-e^2 x^2}\,dx-\frac{e\,(g x)^n}{x^n}\int \frac{x^{n+1}\,(a+c x^2)^p}{d^2-e^2 x^2}\,dx$$

Program code:

```
Int[(g_.*x_)^n_.*(a_+c_.*x_^2)^p_/(d_+e_.*x_),x_Symbol] :=
  d*(g*x)^n/x^n*Int[(x^n*(a+c*x^2)^p)/(d^2-e^2*x^2),x] -
  e*(g*x)^n/x^n*Int[(x^(n+1)*(a+c*x^2)^p)/(d^2-e^2*x^2),x] /;
FreeQ[{a,c,d,e,g,n,p},x] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && Not[IntegersQ[n,2*p]]
```

14: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,(p\in\mathbb{Z}\,\vee\,(m\mid n)\in\mathbb{Z})$

Derivation: Algebraic expansion

Rule 1.2.1.4.14: If $e\,f-d\,g\neq 0\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,(p\in\mathbb{Z}\,\vee\,(m\mid n)\in\mathbb{Z})$**, then**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\int \text{ExpandIntegrand}\left[(d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p,\,x\right]dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)^n*(a+b*x+c*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && (IntegerQ[p] || ILtQ[m,0] && ILtQ[n,0
  Not[IGtQ[m,0] || IGtQ[n,0]]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)^n*(a+c*x^2)^p,x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && NeQ[c*d^2+a*e^2,0] && (IntegerQ[p] || ILtQ[m,0] && ILtQ[n,0]) &&
  Not[IGtQ[m,0] || IGtQ[n,0]]
```

15: $\int (g\,x)^n\,(d+e\,x)^m\,\left(a+c\,x^2\right)^p\,dx$ when $c\,d^2+a\,e^2\neq 0\,\wedge\,m\in\mathbb{Z}^-\,\wedge\,p\notin\mathbb{Z}\,\wedge\,n\notin\mathbb{Z}$

Derivation: Algebraic expansion

Basis: If $m\in\mathbb{Z}$**, then** $(d+e\,x)^m==\left(\frac{d}{d^2-e^2\,x^2}-\frac{e\,x}{d^2-e^2\,x^2}\right)^{-m}$

Note: Resulting integrands are of the form $x^m\,\left(a+b\,x^2\right)^p\,\left(c+d\,x^2\right)^q$ **which are integrable in terms of the Appell hypergeometric function .**

Rule 1.2.1.4.15: If $c\,d^2+a\,e^2\neq 0\,\wedge\,m\in\mathbb{Z}^-\,\wedge\,p\notin\mathbb{Z}\,\wedge\,n\notin\mathbb{Z}$**, then**

$$\int (g\,x)^n\,(d+e\,x)^m\,\left(a+c\,x^2\right)^p\,dx\,\longrightarrow\,\frac{(g\,x)^n}{x^n}\int x^n\,\left(a+c\,x^2\right)^p\,\text{ExpandIntegrand}\left[\left(\frac{d}{d^2-e^2\,x^2}-\frac{e\,x}{d^2-e^2\,x^2}\right)^{-m},\,x\right]dx$$

Program code:

```
Int[(g_.*x_)^n_.*(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (g*x)^n/x^n*Int[ExpandIntegrand[x^n*(a+c*x^2)^p,(d/(d^2-e^2*x^2)-e*x/(d^2-e^2*x^2))^(-m),x],x] /;
FreeQ[{a,c,d,e,g,n,p},x] && NeQ[c*d^2+a*e^2,0] && ILtQ[m,0] && Not[IntegerQ[p]] && Not[IntegerQ[n]]
```

U: $\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$

- **Rule 1.2.1.4.U:**

$$\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow\;\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Unintegrable[(d+e*x)^m*(f+g*x)^n*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && Not[IGtQ[m,0] || IGtQ[n,0]]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)^n_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  Unintegrable[(d+e*x)^m*(f+g*x)^n*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,f,g,m,n,p},x] && Not[IGtQ[m,0] || IGtQ[n,0]]
```

S: $\int (d+e\,u)^m\,(f+g\,u)^n\,\left(a+b\,u+c\,u^2\right)^p\,dx$ when $u == h+j\,x$

- **Derivation: Integration by substitution**
- **Rule 1.2.1.4.S: If** $u == h+j\,x$, **then**

$$\int (d+e\,u)^m\,(f+g\,u)^n\,\left(a+b\,u+c\,u^2\right)^p\,dx\;\longrightarrow\;\frac{1}{j}\,\text{Subst}\left[\int (d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p\,dx,\;x,\;u\right]$$

- **Program code:**

```
Int[(d_.+e_.*u_)^m_.*(f_.+g_.*u_)^n_.*(a_+b_.*u_+c_.*u_^2)^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(d+e*x)^m*(f+g*x)^n*(a+b*x+c*x^2)^p,x],x,u] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && LinearQ[u,x] && NeQ[u,x]
```

```
Int[(d_.+e_.*u_)^m_.*(f_.+g_.*u_)^n_.*(a_+c_.*u_^2)^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(d+e*x)^m*(f+g*x)^n*(a+c*x^2)^p,x],x,u] /;
FreeQ[{a,c,d,e,f,g,m,n,p},x] && LinearQ[u,x] && NeQ[u,x]
```