Rules for integrands of the form $\tan\left[a+b\,x+c\,x^2\right]^n$

X: $\int \tan\left[a+b\,x+c\,x^2\right]^n\,dx$

- **Rule:**

$$\int \tan\left[a+b\,x+c\,x^2\right]^n\,dx \longrightarrow \int \tan\left[a+b\,x+c\,x^2\right]^n\,dx$$

- **Program code:**

```
Int[Tan[a_.+b_.*x_+c_.*x_^2]^n_.,x_Symbol] :=
  Unintegrable[Tan[a+b*x+c*x^2]^n,x] /;
FreeQ[{a,b,c,n},x]
```

```
Int[Cot[a_.+b_.*x_+c_.*x_^2]^n_.,x_Symbol] :=
  Unintegrable[Cot[a+b*x+c*x^2]^n,x] /;
FreeQ[{a,b,c,n},x]
```

Rules for integrands of the form $(d+e\,x)^m\tan\left[a+b\,x+c\,x^2\right]^n$

1. $\int (d+e\,x)\,\tan\left[a+b\,x+c\,x^2\right]\,dx$

1: $\int (d+e\,x)\,\tan\left[a+b\,x+c\,x^2\right]\,dx$ when $2\,c\,d-b\,e==0$

- **Rule: If $2\,c\,d-b\,e==0$, then**

$$\int (d+e\,x)\,\tan\left[a+b\,x+c\,x^2\right]\,dx \longrightarrow -\frac{e\,\mathrm{Log}\left[\cos\left[a+b\,x+c\,x^2\right]\right]}{2\,c}$$

- **Program code:**

```
Int[(d_+e_.*x_)*Tan[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -e*Log[Cos[a+b*x+c*x^2]]/(2*c) /;
FreeQ[{a,b,c,d,e},x] && EqQ[2*c*d-b*e,0]
```

```
Int[(d_+e_.*x_)*Cot[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*Log[Sin[a+b*x+c*x^2]]/(2*c) /;
FreeQ[{a,b,c,d,e},x] && EqQ[2*c*d-b*e,0]
```

2: $\int (d+e\,x)\,\text{Tan}\left[a+b\,x+c\,x^2\right]\,dx$ when $2\,c\,d-b\,e\neq 0$

Rule: If $2\,c\,d-b\,e\neq 0$, then

$$\int (d+e\,x)\,\text{Tan}\left[a+b\,x+c\,x^2\right]\,dx \longrightarrow -\frac{e\,\text{Log}\left[\text{Cos}\left[a+b\,x+c\,x^2\right]\right]}{2\,c}+\frac{2\,c\,d-b\,e}{2\,c}\int \text{Tan}\left[a+b\,x+c\,x^2\right]\,dx$$

Program code:

```
Int[(d_.+e_.*x_)*Tan[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -e*Log[Cos[a+b*x+c*x^2]]/(2*c) +
  (2*c*d-b*e)/(2*c)*Int[Tan[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[2*c*d-b*e,0]
```

```
Int[(d_.+e_.*x_)*Cot[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*Log[Sin[a+b*x+c*x^2]]/(2*c) +
  (2*c*d-b*e)/(2*c)*Int[Cot[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[2*c*d-b*e,0]
```

x: $\int (d+e\,x)^m\,\text{Tan}\left[a+b\,x+c\,x^2\right]\,dx$ when $m>1$

Note: This rule is valid, but to be useful need a rule for reducing integrands of the form $x^m\,\text{Log}\left[\text{Cos}\left[a+b\,x+c\,x^2\right]\right]$.

Rule: If $m>1$, then

$$\int x^m\,\text{Tan}\left[a+b\,x+c\,x^2\right]\,dx \longrightarrow$$
$$-\frac{x^{m-1}\,\text{Log}\left[\text{Cos}\left[a+b\,x+c\,x^2\right]\right]}{2\,c}-\frac{b}{2\,c}\int x^{m-1}\,\text{Tan}\left[a+b\,x+c\,x^2\right]\,dx+\frac{m-1}{2\,c}\int x^{m-2}\,\text{Log}\left[\text{Cos}\left[a+b\,x+c\,x^2\right]\right]\,dx$$

Program code:

```
(* Int[x_^m_*Tan[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -x^(m-1)*Log[Cos[a+b*x+c*x^2]]/(2*c) -
  b/(2*c)*Int[x^(m-1)*Tan[a+b*x+c*x^2],x] +
  (m-1)/(2*c)*Int[x^(m-2)*Log[Cos[a+b*x+c*x^2]],x] /;
FreeQ[{a,b,c},x] && GtQ[m,1] *)
```

```
(* Int[x_^m_*Cot[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  x^(m-1)*Log[Sin[a+b*x+c*x^2]]/(2*c) -
  b/(2*c)*Int[x^(m-1)*Cot[a+b*x+c*x^2],x] -
  (m-1)/(2*c)*Int[x^(m-2)*Log[Sin[a+b*x+c*x^2]],x] /;
FreeQ[{a,b,c},x] && GtQ[m,1]*)
```

X: $\int (d+e\,x)^m\,\mathrm{Tan}\left[a+b\,x+c\,x^2\right]^n\,dx$

- **Rule:**

$$\int (d+e\,x)^m\,\mathrm{Tan}\left[a+b\,x+c\,x^2\right]^n\,dx \longrightarrow \int (d+e\,x)^m\,\mathrm{Tan}\left[a+b\,x+c\,x^2\right]^n\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_.*Tan[a_.+b_.*x_+c_.*x_^2]^n_.,x_Symbol] :=
  Unintegrable[(d+e*x)^m*Tan[a+b*x+c*x^2]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

```
Int[(d_.+e_.*x_)^m_.*Cot[a_.+b_.*x_+c_.*x_^2]^n_.,x_Symbol] :=
  Unintegrable[(d+e*x)^m*Cot[a+b*x+c*x^2]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```