Rules for integrands involving Bessel functions

1. $\int u\,\text{BesselJ}[n,\ a+b\,x]\,dx$

   1. $\int \text{BesselJ}[n,\ a+b\,x]\,dx$

      1. $\int \text{BesselJ}[n,\ a+b\,x]\,dx$ when $\frac{n+1}{2}\in\mathbb{Z}^+$

         1: $\int \text{BesselJ}[1,\ a+b\,x]\,dx$

- **Rule:**

$$\int \text{BesselJ}[1,\ a+b\,x]\,dx \longrightarrow -\frac{\text{BesselJ}[0,\ a+b\,x]}{b}$$

- **Program code:**

```
Int[BesselJ[1,a_.+b_.*x_],x_Symbol] :=
  -BesselJ[0,a+b*x]/b /;
FreeQ[{a,b},x]
```

2: $\int \text{BesselJ}[n,\ a+b\,x]\,dx$ when $\frac{n-1}{2}\in\mathbb{Z}^+$

- **Basis:** $\text{BesselJ}[n,\ a+b\,x] == -\frac{2\,\partial_x \text{BesselJ}[n-1,a+b\,x]}{b} + \text{BesselJ}[n-2,\ a+b\,x]$

- **Rule: If** $\frac{n-1}{2}\in\mathbb{Z}^+$**, then**

$$\int \text{BesselJ}[n,\ a+b\,x]\,dx \longrightarrow -\frac{2\,\text{BesselJ}[n-1,\ a+b\,x]}{b} + \int \text{BesselJ}[n-2,\ a+b\,x]\,dx$$

- **Program code:**

```
Int[BesselJ[n_,a_.+b_.*x_],x_Symbol] :=
  -2*BesselJ[n-1,a+b*x]/b + Int[BesselJ[n-2,a+b*x],x] /;
FreeQ[{a,b},x] && IGtQ[(n-1)/2,0]
```

x: $\int \text{BesselJ}[n,\ a+b\,x]\, dx$ when $n \in \mathbb{Z}^-$

- **Derivation: Algebraic simplification**
- **Basis: If $n \in \mathbb{Z}$, then $\text{BesselJ}[n,\ z] == (-1)^n\, \text{BesselJ}[-n,\ z]$**
- **Note: This rule not necessary since *Mathematica* automatically simplifies $\text{BesselJ}[n,\ a+b\,x]$ to $(-1)^n\, \text{BesselJ}[-n,\ z]$ if $n \in \mathbb{Z}^-$.**
- **Rule: If $n \in \mathbb{Z}^-$, then**

$$\int \text{BesselJ}[n,\ a+b\,x]\, dx \longrightarrow (-1)^n \int \text{BesselJ}[-n,\ a+b\,x]\, dx$$

- **Program code:**

```
(* Int[BesselJ[n_,a_.+b_.*x_],x_Symbol] :=
  (-1)^n*Int[BesselJ[-n,a+b*x],x] /;
FreeQ[{a,b},x] && ILtQ[n,0] *)
```

2: $\int \text{BesselJ}[n,\ a+b\,x]\, dx$

- **Rule:**

$$\int \text{BesselJ}[n,\ a+b\,x]\, dx \longrightarrow \frac{(a+b\,x)^{n+1}\, \text{HypergeometricPFQ}\left[\left\{\frac{n+1}{2}\right\},\ \left\{\frac{n+3}{2},\ n+1\right\},\ -\frac{1}{4}\,(a+b\,x)^2\right]}{2^n\, b\, \text{Gamma}[n+2]}$$

- **Program code:**

```
Int[BesselJ[n_,a_.+b_.*x_],x_Symbol] :=
  (a+b*x)^(n+1)*HypergeometricPFQ[{(n+1)/2},{(n+3)/2,n+1},-1/4*(a+b*x)^2]/(2^n*b*Gamma[n+2]) /;
FreeQ[{a,b,n},x]
```

2. $\int (d\,x)^m\, \text{BesselJ}[n,\ b\,x]\, dx$

3. $\int (c+d\,x)^m\, \text{BesselJ}[n,\ a+b\,x]\, dx$

2. $\int u\, \text{BesselK}[n,\ a+b\,x]\, dx$

3. $\int u\, \text{BesselY}[n,\ a+b\,x]\, dx$