Rules for integrands of the form $(a + b x^n)^p$

0: $\int (b x^n)^p \, dx$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \frac{(b x^n)^p}{x^{n p}} == 0$
- **Basis:** $\frac{(b x^n)^p}{x^{n p}} == \frac{b^{\text{IntPart}[p]} (b x^n)^{\text{FracPart}[p]}}{x^{n \, \text{FracPart}[p]}}$
- **Rule 1.1.3.1.0:**

$$\int (b x^n)^p \, dx \longrightarrow \frac{b^{\text{IntPart}[p]} (b x^n)^{\text{FracPart}[p]}}{x^{n \, \text{FracPart}[p]}} \int x^{n p} \, dx$$

- **Program code:**

```
Int[(b_.*x_^n_)^p_,x_Symbol] :=
  b^IntPart[p]*(b*x^n)^FracPart[p]/x^(n*FracPart[p])*Int[x^(n*p),x] /;
FreeQ[{b,n,p},x]
```

1: $\int (a + b x^n)^p \, dx$ when $n \in \mathbb{F} \bigwedge \frac{1}{n} \in \mathbb{Z}$

- **Derivation: Integration by substitution**
- **Basis: If** $\frac{1}{n} \in \mathbb{Z}$**, then** $F[x^n] == \frac{1}{n} \text{Subst}\left[x^{\frac{1}{n}-1} F[x], x, x^n\right] \partial_x x^n$
- **Rule 1.1.3.1.1: If** $n \in \mathbb{F} \bigwedge \frac{1}{n} \in \mathbb{Z}$**, then**

$$\int (a + b x^n)^p \, dx \longrightarrow \frac{1}{n} \text{Subst}\left[\int x^{\frac{1}{n}-1} (a + b x)^p \, dx, x, x^n\right]$$

- **Program code:**

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  1/n*Subst[Int[x^(1/n-1)*(a+b*x)^p,x],x,x^n] /;
FreeQ[{a,b,p},x] && FractionQ[n] && IntegerQ[1/n]
```

2. $\int (a+b\,x^n)^p\,dx$ when $\frac{1}{n}+p\in\mathbb{Z}^- \bigwedge p\neq -1$

1: $\int (a+b\,x^n)^p\,dx$ when $\frac{1}{n}+p+1==0$

Reference: G&R 2.110.2, CRC 88d with $n\,(p+1)+1==0$

- **Derivation:** Binomial recurrence 3b with $m==0$ and $\frac{1}{n}+p+1==0$
- **Rule 1.1.3.1.2.1:** If $\frac{1}{n}+p+1==0$, then

$$\int (a+b\,x^n)^p\,dx \longrightarrow \frac{x\,(a+b\,x^n)^{p+1}}{a}$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  x*(a+b*x^n)^(p+1)/a /;
FreeQ[{a,b,n,p},x] && EqQ[1/n+p+1,0]
```

2: $\int (a + b x^n)^p \, dx$ when $\frac{1}{n} + p + 1 \in \mathbb{Z}^- \bigwedge p \neq -1$

Reference: G&R 2.110.2, CRC 88d

Derivation: Binomial recurrence 2b

Derivation: Integration by parts

Basis: $x^m (a + b x^n)^p == x^{m+n p+n+1} \frac{(a+b x^n)^p}{x^{n (p+1)+1}}$

Basis: $\int \frac{(a+b x^n)^p}{x^{n (p+1)+1}} \, dx == -\frac{(a+b x^n)^{p+1}}{x^{n (p+1)} a n (p+1)}$

Rule 1.1.3.1.2.2: If $\frac{1}{n} + p + 1 \in \mathbb{Z}^- \bigwedge p \neq -1$**, then**

$$\int (a + b x^n)^p \, dx \longrightarrow -\frac{x (a + b x^n)^{p+1}}{a n (p + 1)} + \frac{n (p + 1) + 1}{a n (p + 1)} \int (a + b x^n)^{p+1} \, dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  -x*(a+b*x^n)^(p+1)/(a*n*(p+1)) +
  (n*(p+1)+1)/(a*n*(p+1))*Int[(a+b*x^n)^(p+1),x] /;
FreeQ[{a,b,n,p},x] && ILtQ[Simplify[1/n+p+1],0] && NeQ[p,-1]
```

3: $\int (a + b x^n)^p \, dx$ when $n < 0 \wedge p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $p \in \mathbb{Z}$**, then** $(a + b x^n)^p == x^{n p} (b + a x^{-n})^p$

Rule 1.1.3.1.3: If $n < 0 \wedge p \in \mathbb{Z}$**, then**

$$\int (a + b x^n)^p \, dx \longrightarrow \int x^{n p} (b + a x^{-n})^p \, dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  Int[x^(n*p)*(b+a*x^(-n))^p,x] /;
FreeQ[{a,b},x] && LtQ[n,0] && IntegerQ[p]
```

4. $\int (a+b\,x^n)^p\,dx$ when $n \in \mathbb{Z}$

1. $\int (a+b\,x^n)^p\,dx$ when $n \in \mathbb{Z}^+$

1. $\int (a+b\,x^n)^p\,dx$ when $n \in \mathbb{Z}^+ \bigwedge p > 0$

1: $\int (a+b\,x^n)^p\,dx$ when $n \in \mathbb{Z}^+ \bigwedge p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.3.1.4.1.1.1: If $n \in \mathbb{Z}^+ \bigwedge p \in \mathbb{Z}^+$, then

$$\int (a+b\,x^n)^p\,dx \longrightarrow \int \text{ExpandIntegrand}\left[(a+b\,x^n)^p,\,x\right]dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x^n)^p,x],x] /;
FreeQ[{a,b},x] && IGtQ[n,0] && IGtQ[p,0]
```

2: $\int (a+b\,x^n)^p\,dx$ when $n\in\mathbb{Z}^+ \land p>0$

Reference: G&R 2.110.1, CRC 88b

Derivation: Binomial recurrence 1b

Derivation: Inverted integration by parts

Note: If $n\in\mathbb{Z}^+ \land p>0$, then $n\,p+1\neq 0$.

Rule 1.1.3.1.4.1.1.2: If $n\in\mathbb{Z}^+ \land p>0$, then

$$\int (a+b\,x^n)^p\,dx \longrightarrow \frac{x\,(a+b\,x^n)^p}{n\,p+1}+\frac{a\,n\,p}{n\,p+1}\int (a+b\,x^n)^{p-1}\,dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  x*(a+b*x^n)^p/(n*p+1) +
  a*n*p/(n*p+1)*Int[(a+b*x^n)^(p-1),x] /;
FreeQ[{a,b},x] && IGtQ[n,0] && GtQ[p,0] &&
  (IntegerQ[2*p] || EqQ[n,2] && IntegerQ[4*p] || EqQ[n,2] && IntegerQ[3*p] || LtQ[Denominator[p+1/n],Denominator[p]])
```

2. $\int (a + b\,x^n)^p \, dx$ when $n \in \mathbb{Z}^+ \bigwedge p < -1$

   1. $\int \frac{1}{\left(a + b\,x^2\right)^{5/4}} \, dx$ when $a \not< 0 \bigwedge \frac{b}{a} > 0$

      1: $\int \frac{1}{\left(a + b\,x^2\right)^{5/4}} \, dx$ when $a > 0 \bigwedge \frac{b}{a} > 0$

– **Contributed by Martin Welz on 7 August 2016**

- **Rule 1.1.3.1.4.1.2.1.1: If $a > 0 \bigwedge \frac{b}{a} > 0$, then**

$$\int \frac{1}{\left(a + b\,x^2\right)^{5/4}} \, dx \longrightarrow \frac{2}{a^{5/4}\sqrt{\frac{b}{a}}} \,\text{EllipticE}\left[\frac{1}{2}\,\text{ArcTan}\left[\sqrt{\frac{b}{a}}\;x\right],\,2\right]$$

– **Program code:**

```
Int[1/(a_+b_.*x_^2)^(5/4),x_Symbol] :=
  2/(a^(5/4)*Rt[b/a,2])*EllipticE[1/2*ArcTan[Rt[b/a,2]*x],2] /;
FreeQ[{a,b},x] && GtQ[a,0] && PosQ[b/a]
```

2: $\int \frac{1}{\left(a+b\,x^2\right)^{5/4}}\,dx$ when $a \not< 0 \bigwedge \frac{b}{a}>0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\left(1+\frac{b\,x^2}{a}\right)^{1/4}}{\left(a+b\,x^2\right)^{1/4}} == 0$
- **Rule 1.1.3.1.4.1.2.1.2: If** $a \not< 0 \bigwedge \frac{b}{a}>0$**, then**

$$\int \frac{1}{\left(a+b\,x^2\right)^{5/4}}\,dx \longrightarrow \frac{\left(1+\frac{b\,x^2}{a}\right)^{1/4}}{a\left(a+b\,x^2\right)^{1/4}} \int \frac{1}{\left(1+\frac{b\,x^2}{a}\right)^{5/4}}\,dx$$

Program code:

```
Int[1/(a_+b_.*x_^2)^(5/4),x_Symbol] :=
  (1+b*x^2/a)^(1/4)/(a*(a+b*x^2)^(1/4))*Int[1/(1+b*x^2/a)^(5/4),x] /;
FreeQ[{a,b},x] && PosQ[a] && PosQ[b/a]
```

2: $\int \frac{1}{\left(a+b x^2\right)^{7/6}} \, dx$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x \frac{1}{(a+b x^2)^{2/3} \left(\frac{a}{a+b x^2}\right)^{2/3}} = 0$

- **Basis:** $\frac{\left(\frac{a}{a+b x^2}\right)^{2/3}}{\sqrt{a+b x^2}} = \text{Subst}\left[\frac{1}{(1-b x^2)^{1/3}}, x, \frac{x}{\sqrt{a+b x^2}}\right] \partial_x \frac{x}{\sqrt{a+b x^2}}$

Rule 1.1.3.1.4.1.2.2:

$$\int \frac{1}{\left(a+b x^2\right)^{7/6}} \, dx \to \frac{1}{\left(a+b x^2\right)^{2/3} \left(\frac{a}{a+b x^2}\right)^{2/3}} \int \frac{\left(\frac{a}{a+b x^2}\right)^{2/3}}{\sqrt{a+b x^2}} \, dx$$

$$\to \frac{1}{\left(a+b x^2\right)^{2/3} \left(\frac{a}{a+b x^2}\right)^{2/3}} \text{Subst}\left[\int \frac{1}{\left(1-b x^2\right)^{1/3}} \, dx, x, \frac{x}{\sqrt{a+b x^2}}\right]$$

- **Program code:**

```
Int[1/(a_+b_.*x_^2)^(7/6),x_Symbol] :=
  1/((a +b*x^2)^(2/3)*(a/(a+b*x^2))^(2/3))*Subst[Int[1/(1-b*x^2)^(1/3),x],x,x/Sqrt[a+b*x^2]] /;
FreeQ[{a,b},x]
```

3: $\int (a+b x^n)^p \, dx$ when $n \in \mathbb{Z}^+ \land p < -1$

Reference: G&R 2.110.2, CRC 88d

Derivation: Binomial recurrence 2b

Derivation: Integration by parts

- **Basis:** $(a+b x^n)^p == x^{n(p+1)+1} \frac{(a+b x^n)^p}{x^{n(p+1)+1}}$
- **Basis:** $\int \frac{(a+b x^n)^p}{x^{n(p+1)+1}} \, dx == -\frac{(a+b x^n)^{p+1}}{x^{n(p+1)} \, a \, n \, (p+1)}$

Rule 1.1.3.1.4.1.2.3: If $n \in \mathbb{Z}^+ \land p < -1$**, then**

$$\int (a+b x^n)^p \, dx \longrightarrow -\frac{x (a+b x^n)^{p+1}}{a \, n \, (p+1)} + \frac{n (p+1) + 1}{a \, n \, (p+1)} \int (a+b x^n)^{p+1} \, dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  -x*(a+b*x^n)^(p+1)/(a*n*(p+1)) +
  (n*(p+1)+1)/(a*n*(p+1))*Int[(a+b*x^n)^(p+1),x] /;
FreeQ[{a,b},x] && IGtQ[n,0] && LtQ[p,-1] &&
  (IntegerQ[2*p] || n==2 && IntegerQ[4*p] || n==2 && IntegerQ[3*p] || Denominator[p+1/n]<Denominator[p])
```

3. $\int \frac{1}{a + b x^n} \, dx$ when $n \in \mathbb{Z}^+$

1. $\int \frac{1}{a + b x^n} \, dx$ when $\frac{n-1}{2} \in \mathbb{Z}^+$

1: $\int \frac{1}{a + b x^3} \, dx$

Reference: G&R 2.126.1.2, CRC 74

Derivation: Algebraic expansion

- **Basis:** $\frac{1}{a+b x^3} == \frac{1}{3 a^{2/3} \left(a^{1/3}+b^{1/3} x\right)} + \frac{2 a^{1/3}-b^{1/3} x}{3 a^{2/3} \left(a^{2/3}-a^{1/3} b^{1/3} x+b^{2/3} x^2\right)}$

Rule 1.1.3.1.4.1.3.1.1:

$$\int \frac{1}{a + b x^3} \, dx \longrightarrow \frac{1}{3 a^{2/3}} \int \frac{1}{a^{1/3} + b^{1/3} x} \, dx + \frac{1}{3 a^{2/3}} \int \frac{2 a^{1/3} - b^{1/3} x}{a^{2/3} - a^{1/3} b^{1/3} x + b^{2/3} x^2} \, dx$$

Program code:

```
Int[1/(a_+b_.*x_^3),x_Symbol] :=
  1/(3*Rt[a,3]^2)*Int[1/(Rt[a,3]+Rt[b,3]*x),x] +
  1/(3*Rt[a,3]^2)*Int[(2*Rt[a,3]-Rt[b,3]*x)/(Rt[a,3]^2-Rt[a,3]*Rt[b,3]*x+Rt[b,3]^2*x^2),x] /;
FreeQ[{a,b},x]
```

x. $\int \frac{1}{a + b x^5} \, dx$

1: $\int \frac{1}{a + b x^5} \, dx$ when $\frac{a}{b} > 0$

Derivation: Algebraic expansion

- **Basis: If** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/5}$**, then** $\frac{1}{a+b x^5} == \frac{r}{5 a (r+s x)} + \frac{2 r \left(r-\frac{1}{4} \left(1-\sqrt{5}\right) s x\right)}{5 a \left(r^2-\frac{1}{2} \left(1-\sqrt{5}\right) r s x+s^2 x^2\right)} + \frac{2 r \left(r-\frac{1}{4} \left(1+\sqrt{5}\right) s x\right)}{5 a \left(r^2-\frac{1}{2} \left(1+\sqrt{5}\right) r s x+s^2 x^2\right)}$

- **Note: This rule not necessary for host systems that automatically simplify** $\text{Cos}\left[\frac{k \pi}{5}\right]$ **to radicals when** k **is an integer.**

- **Rule 1.1.3.1.4.1.3.1.2.1: If** $\frac{a}{b} > 0$**, let** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/5}$**, then**

$$\int \frac{1}{a+b\,x^5}\,dx \to \frac{r}{5\,a}\int \frac{1}{r+s\,x}\,dx+\frac{2\,r}{5\,a}\int \frac{r-\frac{1}{4}\left(1-\sqrt{5}\right)s\,x}{r^2-\frac{1}{2}\left(1-\sqrt{5}\right)r\,s\,x+s^2\,x^2}\,dx+\frac{2\,r}{5\,a}\int \frac{r-\frac{1}{4}\left(1+\sqrt{5}\right)s\,x}{r^2-\frac{1}{2}\left(1+\sqrt{5}\right)r\,s\,x+s^2\,x^2}\,dx$$

Program code:

```
(* Int[1/(a_+b_.*x_^5),x_Symbol] :=
  With[{r=Numerator[Rt[a/b,5]], s=Denominator[Rt[a/b,5]]},
  r/(5*a)*Int[1/(r+s*x),x] +
   2*r/(5*a)*Int[(r-1/4*(1-Sqrt[5])*s*x)/(r^2-1/2*(1-Sqrt[5])*r*s*x+s^2*x^2),x] +
  2*r/(5*a)*Int[(r-1/4*(1+Sqrt[5])*s*x)/(r^2-1/2*(1+Sqrt[5])*r*s*x+s^2*x^2),x]] /;
FreeQ[{a,b},x] && PosQ[a/b] *)
```

2: $\int \frac{1}{a+b\,x^5}\,dx$ when $\frac{a}{b} \ngtr 0$

Derivation: Algebraic expansion

- **Basis: If** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/5}$**, then** $\frac{1}{a+b\,x^5} == \frac{r}{5\,a\,(r-s\,x)}+\frac{2\,r\left(r+\frac{1}{4}\left(1-\sqrt{5}\right)s\,x\right)}{5\,a\left(r^2+\frac{1}{2}\left(1-\sqrt{5}\right)r\,s\,x+s^2\,x^2\right)}+\frac{2\,r\left(r+\frac{1}{4}\left(1+\sqrt{5}\right)s\,x\right)}{5\,a\left(r^2+\frac{1}{2}\left(1+\sqrt{5}\right)r\,s\,x+s^2\,x^2\right)}$
- **Note: This rule not necessary for host systems that automatically simplify** $\text{Cos}\left[\frac{k\,\pi}{5}\right]$ **to radicals when** k **is an integer.**
- **Rule 1.1.3.1.4.1.3.1.2.2: If** $\frac{a}{b} \ngtr 0$**, let** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/5}$**, then**

$$\int \frac{1}{a+b\,x^5}\,dx \to \frac{r}{5\,a}\int \frac{1}{r-s\,x}\,dx+\frac{2\,r}{5\,a}\int \frac{r+\frac{1}{4}\left(1-\sqrt{5}\right)s\,x}{r^2+\frac{1}{2}\left(1-\sqrt{5}\right)r\,s\,x+s^2\,x^2}\,dx+\frac{2\,r}{5\,a}\int \frac{r+\frac{1}{4}\left(1+\sqrt{5}\right)s\,x}{r^2+\frac{1}{2}\left(1+\sqrt{5}\right)r\,s\,x+s^2\,x^2}\,dx$$

Program code:

```
(* Int[1/(a_+b_.*x_^5),x_Symbol] :=
  With[{r=Numerator[Rt[-a/b,5]], s=Denominator[Rt[-a/b,5]]},
  r/(5*a)*Int[1/(r-s*x),x] +
   2*r/(5*a)*Int[(r+1/4*(1-Sqrt[5])*s*x)/(r^2+1/2*(1-Sqrt[5])*r*s*x+s^2*x^2),x] +
  2*r/(5*a)*Int[(r+1/4*(1+Sqrt[5])*s*x)/(r^2+1/2*(1+Sqrt[5])*r*s*x+s^2*x^2),x]] /;
FreeQ[{a,b},x] && NegQ[a/b] *)
```

3. $\int \frac{1}{a+b\,x^n}\,dx$ when $\frac{n-3}{2} \in \mathbb{Z}^+$

1: $\int \frac{1}{a+b x^n} \, dx$ when $\frac{n-3}{2} \in \mathbb{Z}^+ \bigwedge \frac{a}{b} > 0$

Derivation: Algebraic expansion

- **Basis: If** $\frac{n-1}{2} \in \mathbb{Z}$ **and** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/n}$**, then** $\frac{1}{a+b z^n} == \frac{r}{a n (r+s z)} + \frac{2 r}{a n} \sum_{k=1}^{\frac{n-1}{2}} \frac{r-s \cos\left[\frac{(2k-1)\pi}{n}\right] z}{r^2-2 r s \cos\left[\frac{(2k-1)\pi}{n}\right] z+s^2 z^2}$
- **Rule 1.1.3.1.4.1.3.1.3.1: If** $\frac{n-3}{2} \in \mathbb{Z}^+ \bigwedge \frac{a}{b} > 0$**, let** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/n}$**, then**

$$\int \frac{1}{a+b x^n} \, dx \to \frac{r}{a n} \int \frac{1}{r+s x} \, dx + \frac{2 r}{a n} \sum_{k=1}^{\frac{n-1}{2}} \int \frac{r-s \cos\left[\frac{(2k-1)\pi}{n}\right] x}{r^2-2 r s \cos\left[\frac{(2k-1)\pi}{n}\right] x+s^2 x^2} \, dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[a/b,n]], s=Denominator[Rt[a/b,n]], k, u},
  u=Int[(r-s*Cos[(2*k-1)*Pi/n]*x)/(r^2-2*r*s*Cos[(2*k-1)*Pi/n]*x+s^2*x^2),x];
  r/(a*n)*Int[1/(r+s*x),x] + Dist[2*r/(a*n),Sum[u,{k,1,(n-1)/2}],x]] /;
FreeQ[{a,b},x] && IGtQ[(n-3)/2,0] && PosQ[a/b]
```

2: $\int \frac{1}{a+b x^n} \, dx$ when $\frac{n-3}{2} \in \mathbb{Z}^+ \bigwedge \frac{a}{b} \not> 0$

Derivation: Algebraic expansion

- **Basis: If** $\frac{n-1}{2} \in \mathbb{Z}$ **and** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/n}$**, then** $\frac{1}{a+b z^n} == \frac{r}{a n (r-s z)} + \frac{2 r}{a n} \sum_{k=1}^{\frac{n-1}{2}} \frac{r+s \cos\left[\frac{(2k-1)\pi}{n}\right] z}{r^2+2 r s \cos\left[\frac{(2k-1)\pi}{n}\right] z+s^2 z^2}$
- **Rule 1.1.3.1.4.1.3.1.3.2: If** $\frac{n-3}{2} \in \mathbb{Z}^+ \bigwedge \frac{a}{b} \not> 0$**, let** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/n}$**, then**

$$\int \frac{1}{a+b x^n} \, dx \to \frac{r}{a n} \int \frac{1}{r-s x} \, dx + \frac{2 r}{a n} \sum_{k=1}^{\frac{n-1}{2}} \int \frac{r+s \cos\left[\frac{(2k-1)\pi}{n}\right] x}{r^2+2 r s \cos\left[\frac{(2k-1)\pi}{n}\right] x+s^2 x^2} \, dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[-a/b,n]], s=Denominator[Rt[-a/b,n]], k, u},
  u=Int[(r+s*Cos[(2*k-1)*Pi/n]*x)/(r^2+2*r*s*Cos[(2*k-1)*Pi/n]*x+s^2*x^2),x];
  r/(a*n)*Int[1/(r-s*x),x] + Dist[2*r/(a*n),Sum[u,{k,1,(n-1)/2}],x]] /;
FreeQ[{a,b},x] && IGtQ[(n-3)/2,0] && NegQ[a/b]
```

2. $\int \frac{1}{a+b\,x^n}\,dx$ when $\frac{n}{2}\in\mathbb{Z}^+$

1. $\int \frac{1}{a+b\,x^n}\,dx$ when $\frac{n+2}{4}\in\mathbb{Z}^+$

1. $\int \frac{1}{a+b\,x^2}\,dx$

1: $\int \frac{1}{a+b\,x^2}\,dx$ when $\frac{a}{b}>0$

Reference: G&R 2.124.1a, CRC 60, A&S 3.3.21

Derivation: Primitive rule

Basis: $\text{ArcTan}'[z] == \frac{1}{1+z^2}$

Rule 1.1.3.1.4.1.3.2.1.1.1: If $\frac{a}{b}>0$**, then**

$$\int \frac{1}{a+b\,x^2}\,dx \to \frac{\sqrt{\frac{a}{b}}}{a}\,\text{ArcTan}\left[\frac{x}{\sqrt{\frac{a}{b}}}\right]$$

Program code:

```
Int[1/(a_+b_.*x_^2),x_Symbol] :=
  1/(Rt[a,2]*Rt[b,2])*ArcTan[Rt[b,2]*x/Rt[a,2]] /;
FreeQ[{a,b},x] && PosQ[a/b] && (GtQ[a,0] || GtQ[b,0])
```

```
Int[1/(a_+b_.*x_^2),x_Symbol] :=
  -1/(Rt[-a,2]*Rt[-b,2])*ArcTan[Rt[-b,2]*x/Rt[-a,2]] /;
FreeQ[{a,b},x] && PosQ[a/b] && (LtQ[a,0] || LtQ[b,0])
```

```
Int[1/(a_+b_.*x_^2),x_Symbol] :=
(*Rt[b/a,2]/b*ArcTan[Rt[b/a,2]*x] /; *)
  Rt[a/b,2]/a*ArcTan[x/Rt[a/b,2]] /;
FreeQ[{a,b},x] && PosQ[a/b]
```

2: $\int \frac{1}{a+b\,x^2}\,dx$ when $\frac{a}{b} \not> 0$

Reference: G&R 2.124.1b', CRC 61b, A&S 3.3.23

Derivation: Primitive rule

Basis: $\text{ArcTanh}'[z] == \frac{1}{1-z^2}$

Rule 1.1.3.1.4.1.3.2.1.1.2: If $\frac{a}{b} \not> 0$**, then**

$$\int \frac{1}{a+b\,x^2}\,dx \to \frac{\sqrt{-\frac{a}{b}}}{a}\,\text{ArcTanh}\left[\frac{x}{\sqrt{-\frac{a}{b}}}\right]$$

Program code:

```
Int[1/(a_+b_.*x_^2),x_Symbol] :=
  1/(Rt[a,2]*Rt[-b,2])*ArcTanh[Rt[-b,2]*x/Rt[a,2]] /;
FreeQ[{a,b},x] && NegQ[a/b] && (GtQ[a,0] || LtQ[b,0])
```

```
Int[1/(a_+b_.*x_^2),x_Symbol] :=
  -1/(Rt[-a,2]*Rt[b,2])*ArcTanh[Rt[b,2]*x/Rt[-a,2]] /;
FreeQ[{a,b},x] && NegQ[a/b] && (LtQ[a,0] || GtQ[b,0])
```

```
Int[1/(a_+b_.*x_^2),x_Symbol] :=
(*-Rt[-b/a,2]/b*ArcTanh[Rt[-b/a,2]*x] /; *)
  Rt[-a/b,2]/a*ArcTanh[x/Rt[-a/b,2]] /;
FreeQ[{a,b},x] && NegQ[a/b]
```

2. $\int \frac{1}{a+b\,x^n}\,dx$ when $\frac{n-2}{4}\in\mathbb{Z}^+$

1: $\int \frac{1}{a+b\,x^n}\,dx$ when $\frac{n-2}{4}\in\mathbb{Z}^+ \bigwedge \frac{a}{b}>0$

Derivation: Algebraic expansion

- **Basis: If** $\frac{n-2}{4}\in\mathbb{Z}$ **and** $\frac{r}{s}=\left(\frac{a}{b}\right)^{1/n}$**, then** $\frac{1}{a+b\,z^n}==\frac{2\,r^2}{a\,n\,(r^2+s^2\,z^2)}+\frac{4\,r^2}{a\,n}\sum_{k=1}^{\frac{n-2}{4}}\frac{r^2-s^2\,\mathrm{Cos}\left[\frac{2\,(2\,k-1)\,\pi}{n}\right]z^2}{r^4-2\,r^2\,s^2\,\mathrm{Cos}\left[\frac{2\,(2\,k-1)\,\pi}{n}\right]z^2+s^4\,z^4}$
- **Basis:** $\frac{r^2-s^2\,\mathrm{Cos}[2\,\theta]\,z^2}{r^4-2\,r^2\,s^2\,\mathrm{Cos}[2\,\theta]\,z^2+s^4\,z^4}==\frac{1}{2\,r}\left(\frac{r-s\,\mathrm{Cos}[\theta]\,z}{r^2-2\,r\,s\,\mathrm{Cos}[\theta]\,z+s^2\,z^2}+\frac{r+s\,\mathrm{Cos}[\theta]\,z}{r^2+2\,r\,s\,\mathrm{Cos}[\theta]\,z+s^2\,z^2}\right)$
- **Rule 1.1.3.1.4.1.3.2.1.2.1: If** $\frac{n-2}{4}\in\mathbb{Z}^+ \bigwedge \frac{a}{b}>0$**, let** $\frac{r}{s}=\left(\frac{a}{b}\right)^{1/n}$**, then**

$$\int\frac{1}{a+b\,x^n}\,dx\;\longrightarrow\;\frac{2\,r^2}{a\,n}\int\frac{1}{r^2+s^2\,x^2}\,dx+\frac{4\,r^2}{a\,n}\sum_{k=1}^{\frac{n-2}{4}}\int\frac{r^2-s^2\,\mathrm{Cos}\left[\frac{2\,(2\,k-1)\,\pi}{n}\right]x^2}{r^4-2\,r^2\,s^2\,\mathrm{Cos}\left[\frac{2\,(2\,k-1)\,\pi}{n}\right]x^2+s^4\,x^4}\,dx$$

$$\longrightarrow\;\frac{2\,r^2}{a\,n}\int\frac{1}{r^2+s^2\,x^2}\,dx+\frac{2\,r}{a\,n}\sum_{k=1}^{\frac{n-2}{4}}\left(\int\frac{r-s\,\mathrm{Cos}\left[\frac{(2\,k-1)\,\pi}{n}\right]x}{r^2-2\,r\,s\,\mathrm{Cos}\left[\frac{(2\,k-1)\,\pi}{n}\right]x+s^2\,x^2}\,dx+\int\frac{r+s\,\mathrm{Cos}\left[\frac{(2\,k-1)\,\pi}{n}\right]x}{r^2+2\,r\,s\,\mathrm{Cos}\left[\frac{(2\,k-1)\,\pi}{n}\right]x+s^2\,x^2}\,dx\right)$$

Program code:

```
Int[1/(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[a/b,n]], s=Denominator[Rt[a/b,n]], k, u, v},
  u=Int[(r-s*Cos[(2*k-1)*Pi/n]*x)/(r^2-2*r*s*Cos[(2*k-1)*Pi/n]*x+s^2*x^2),x] +
    Int[(r+s*Cos[(2*k-1)*Pi/n]*x)/(r^2+2*r*s*Cos[(2*k-1)*Pi/n]*x+s^2*x^2),x];
  2*r^2/(a*n)*Int[1/(r^2+s^2*x^2),x] + Dist[2*r/(a*n),Sum[u,{k,1,(n-2)/4}],x]] /;
FreeQ[{a,b},x] && IGtQ[(n-2)/4,0] && PosQ[a/b]
```

2: $\int \frac{1}{a+b\,x^n}\,dx$ when $\frac{n-2}{4} \in \mathbb{Z}^+ \bigwedge \frac{a}{b} \not> 0$

Derivation: Algebraic expansion

- **Basis: If** $\frac{n-2}{4} \in \mathbb{Z}$ **and** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/n}$**, then** $\frac{1}{a+b\,z^n} == \frac{2\,r^2}{a\,n\,(r^2-s^2\,z^2)} + \frac{4\,r^2}{a\,n}\sum_{k=1}^{\frac{n-2}{4}} \frac{r^2-s^2\,\text{Cos}\left[\frac{4\,k\,\pi}{n}\right]\,z^2}{r^4-2\,r^2\,s^2\,\text{Cos}\left[\frac{4\,k\,\pi}{n}\right]\,z^2+s^4\,z^4}$

- **Basis:** $\frac{r^2-s^2\,\text{Cos}[2\,\theta]\,z^2}{r^4-2\,r^2\,s^2\,\text{Cos}[2\,\theta]\,z^2+s^4\,z^4} == \frac{1}{2\,r}\left(\frac{r-s\,\text{Cos}[\theta]\,z}{r^2-2\,r\,s\,\text{Cos}[\theta]\,z+s^2\,z^2} + \frac{r+s\,\text{Cos}[\theta]\,z}{r^2+2\,r\,s\,\text{Cos}[\theta]\,z+s^2\,z^2}\right)$

- **Rule 1.1.3.1.4.1.3.2.1.2.2: If** $\frac{n-2}{4} \in \mathbb{Z}^+ \bigwedge \frac{a}{b} \not> 0$**, let** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/n}$**, then**

$$\int \frac{1}{a+b\,x^n}\,dx \longrightarrow \frac{2\,r^2}{a\,n}\int \frac{1}{r^2-s^2\,x^2}\,dx + \frac{4\,r^2}{a\,n}\sum_{k=1}^{\frac{n-2}{4}}\int \frac{r^2-s^2\,\text{Cos}\left[\frac{4\,k\,\pi}{n}\right]\,x^2}{r^4-2\,r^2\,s^2\,\text{Cos}\left[\frac{4\,k\,\pi}{n}\right]\,x^2+s^4\,x^4}\,dx$$

$$\longrightarrow \frac{2\,r^2}{a\,n}\int \frac{1}{r^2-s^2\,x^2}\,dx + \frac{2\,r}{a\,n}\sum_{k=1}^{\frac{n-2}{4}}\left(\int \frac{r-s\,\text{Cos}\left[\frac{2\,k\,\pi}{n}\right]\,x}{r^2-2\,r\,s\,\text{Cos}\left[\frac{2\,k\,\pi}{n}\right]\,x+s^2\,x^2}\,dx + \int \frac{r+s\,\text{Cos}\left[\frac{2\,k\,\pi}{n}\right]\,x}{r^2+2\,r\,s\,\text{Cos}\left[\frac{2\,k\,\pi}{n}\right]\,x+s^2\,x^2}\,dx\right)$$

- **Program code:**

```
Int[1/(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[-a/b,n]], s=Denominator[Rt[-a/b,n]], k, u},
  u=Int[(r-s*Cos[(2*k*Pi)/n]*x)/(r^2-2*r*s*Cos[(2*k*Pi)/n]*x+s^2*x^2),x] +
    Int[(r+s*Cos[(2*k*Pi)/n]*x)/(r^2+2*r*s*Cos[(2*k*Pi)/n]*x+s^2*x^2),x];
  2*r^2/(a*n)*Int[1/(r^2-s^2*x^2),x] + Dist[2*r/(a*n),Sum[u,{k,1,(n-2)/4}],x]] /;
FreeQ[{a,b},x] && IGtQ[(n-2)/4,0] && NegQ[a/b]
```

2. $\int \frac{1}{a+b\,x^n}\,dx$ when $\frac{n}{4}\in\mathbb{Z}^+$

   1. $\int \frac{1}{a+b\,x^4}\,dx$

      1: $\int \frac{1}{a+b\,x^4}\,dx$ when $\frac{a}{b}>0$

Derivation: Algebraic expansion

- **Basis: If** $\frac{r}{s}=\sqrt{\frac{a}{b}}$**, then** $\frac{1}{a+b\,x^4}==\frac{r-s\,x^2}{2\,r\,(a+b\,x^4)}+\frac{r+s\,x^2}{2\,r\,(a+b\,x^4)}$
- **Note: Resulting integrands are of the form** $\frac{d+e\,x^2}{a+c\,x^4}$ **where** $c\,d^2-a\,e^2==0$ **as required by the algebraic trinomial rules.**
- **Rule 1.1.3.1.4.1.3.2.2.1.1: If** $\frac{a}{b}>0$**, let** $\frac{r}{s}=\sqrt{\frac{a}{b}}$**, then**

$$\int \frac{1}{a+b\,x^4}\,dx \longrightarrow \frac{1}{2\,r}\int \frac{r-s\,x^2}{a+b\,x^4}\,dx+\frac{1}{2\,r}\int \frac{r+s\,x^2}{a+b\,x^4}\,dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^4),x_Symbol] :=
  With[{r=Numerator[Rt[a/b,2]], s=Denominator[Rt[a/b,2]]},
  1/(2*r)*Int[(r-s*x^2)/(a+b*x^4),x] + 1/(2*r)*Int[(r+s*x^2)/(a+b*x^4),x]] /;
FreeQ[{a,b},x] && (GtQ[a/b,0] || PosQ[a/b] && AtomQ[SplitProduct[SumBaseQ,a]] && AtomQ[SplitProduct[SumBaseQ,b]])
```

$$2: \int \frac{1}{a+b x^4} \, dx \text{ when } \frac{a}{b} \not> 0$$

Reference: G&R 2.132.1.2', CRC 78'

Derivation: Algebraic expansion

- **Basis: Let** $\frac{r}{s} = \sqrt{-\frac{a}{b}}$**, then** $\frac{1}{a+b z^2} == \frac{r}{2 a (r-s z)} + \frac{r}{2 a (r+s z)}$
- **Rule 1.1.3.1.4.1.3.2.2.1.2: If** $\frac{a}{b} \not> 0$**, let** $\frac{r}{s} = \sqrt{-\frac{a}{b}}$**, then**

$$\int \frac{1}{a+b x^4} \, dx \longrightarrow \frac{r}{2 a} \int \frac{1}{r-s x^2} \, dx + \frac{r}{2 a} \int \frac{1}{r+s x^2} \, dx$$

Program code:

```
Int[1/(a_+b_.*x_^4),x_Symbol] :=
  With[{r=Numerator[Rt[-a/b,2]], s=Denominator[Rt[-a/b,2]]},
  r/(2*a)*Int[1/(r-s*x^2),x] + r/(2*a)*Int[1/(r+s*x^2),x]] /;
FreeQ[{a,b},x] && Not[GtQ[a/b,0]]
```

2. $\int \frac{1}{a+b\,x^n}\,dx$ when $\frac{n}{4}-1\in\mathbb{Z}^+$

1: $\int \frac{1}{a+b\,x^n}\,dx$ when $\frac{n}{4}-1\in\mathbb{Z}^+ \bigwedge \frac{a}{b}>0$

Reference: G&R 2.132.1.1', CRC 77'

Derivation: Algebraic expansion

- Basis: If $\frac{r}{s}=\left(\frac{a}{b}\right)^{1/4}$, then $\frac{1}{a+b\,z^4}==\frac{r\left(\sqrt{2}\,r-s\,z\right)}{2\,\sqrt{2}\,a\left(r^2-\sqrt{2}\,r\,s\,z+s^2\,z^2\right)}+\frac{r\left(\sqrt{2}\,r+s\,z\right)}{2\,\sqrt{2}\,a\left(r^2+\sqrt{2}\,r\,s\,z+s^2\,z^2\right)}$

- Rule 1.1.3.1.4.1.3.2.2.2.1: If $\frac{n}{4}\in\mathbb{Z}^+ \bigwedge n>4 \bigwedge \frac{a}{b}>0$, let $\frac{r}{s}=\left(\frac{a}{b}\right)^{1/4}$, then

$$\int \frac{1}{a+b\,x^n}\,dx \longrightarrow \frac{r}{2\,\sqrt{2}\,a}\int \frac{\sqrt{2}\,r-s\,x^{n/4}}{r^2-\sqrt{2}\,r\,s\,x^{n/4}+s^2\,x^{n/2}}\,dx+\frac{r}{2\,\sqrt{2}\,a}\int \frac{\sqrt{2}\,r+s\,x^{n/4}}{r^2+\sqrt{2}\,r\,s\,x^{n/4}+s^2\,x^{n/2}}\,dx$$

Program code:

```
Int[1/(a_+b_.*x_^n_),x_Symbol] :=
  With[{r=Numerator[Rt[a/b,4]], s=Denominator[Rt[a/b,4]]},
  r/(2*Sqrt[2]*a)*Int[(Sqrt[2]*r-s*x^(n/4))/(r^2-Sqrt[2]*r*s*x^(n/4)+s^2*x^(n/2)),x] +
  r/(2*Sqrt[2]*a)*Int[(Sqrt[2]*r+s*x^(n/4))/(r^2+Sqrt[2]*r*s*x^(n/4)+s^2*x^(n/2)),x]] /;
FreeQ[{a,b},x] && IGtQ[n/4,1] && GtQ[a/b,0]
```

2: $\int \frac{1}{a+b\,x^n}\,dx$ when $\frac{n}{4}-1\in\mathbb{Z}^+ \bigwedge \frac{a}{b} \ngtr 0$

Reference: G&R 2.132.1.2', CRC 78'

Derivation: Algebraic expansion

- **Basis: Let** $\frac{r}{s}=\sqrt{-\frac{a}{b}}$, **then** $\frac{1}{a+b\,z^2} == \frac{r}{2\,a\,(r-s\,z)}+\frac{r}{2\,a\,(r+s\,z)}$

- **Rule 1.1.3.1.4.1.3.2.2.2.2: If** $\frac{n}{4}\in\mathbb{Z}^+ \bigwedge \frac{a}{b}\ngtr 0$**, let** $\frac{r}{s}=\sqrt{-\frac{a}{b}}$**, then**

$$\int \frac{1}{a+b\,x^n}\,dx \longrightarrow \frac{r}{2\,a}\int \frac{1}{r-s\,x^{n/2}}\,dx+\frac{r}{2\,a}\int \frac{1}{r+s\,x^{n/2}}\,dx$$

Program code:

```
Int[1/(a_+b_.*x_^n_),x_Symbol] :=
  With[{r=Numerator[Rt[-a/b,2]], s=Denominator[Rt[-a/b,2]]},
  r/(2*a)*Int[1/(r-s*x^(n/2)),x] + r/(2*a)*Int[1/(r+s*x^(n/2)),x]] /;
FreeQ[{a,b},x] && IGtQ[n/4,1] && Not[GtQ[a/b,0]]
```

4. $\int \frac{1}{\sqrt{a+b\,x^n}}\,dx$ when $n\in\mathbb{Z}^+$

   1. $\int \frac{1}{\sqrt{a+b\,x^2}}\,dx$

      1. $\int \frac{1}{\sqrt{a+b\,x^2}}\,dx$ when $a>0$

         1: $\int \frac{1}{\sqrt{a+b\,x^2}}\,dx$ when $a>0 \bigwedge b>0$

Reference: CRC 278

Derivation: Primitive rule

- **Basis:** $\text{ArcSinh}'[z] == \frac{1}{\sqrt{1+z^2}}$

Rule 1.1.3.1.4.1.4.1.1.1: If $a>0 \land b>0$**, then**

$$\int \frac{1}{\sqrt{a+b\,x^2}}\,dx \to \frac{1}{\sqrt{b}}\,\text{ArcSinh}\Big[\frac{\sqrt{b}\,x}{\sqrt{a}}\Big]$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^2],x_Symbol] :=
  ArcSinh[Rt[b,2]*x/Sqrt[a]]/Rt[b,2] /;
FreeQ[{a,b},x] && GtQ[a,0] && PosQ[b]
```

2: $\int \frac{1}{\sqrt{a+b\,x^2}}\,dx$ when $a>0 \wedge b \ngtr 0$

Reference: G&R 2.271.4b, CRC 279, A&S 3.3.44

Derivation: Primitive rule

Basis: $\text{ArcSin}'[z] = \frac{1}{\sqrt{1-z^2}}$

Rule 1.1.3.1.4.1.4.1.1.2: If $a>0 \wedge b \ngtr 0$, then

$$\int \frac{1}{\sqrt{a+b\,x^2}}\,dx \to \frac{1}{\sqrt{-b}}\,\text{ArcSin}\Big[\frac{\sqrt{-b}\,x}{\sqrt{a}}\Big]$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^2],x_Symbol] :=
  ArcSin[Rt[-b,2]*x/Sqrt[a]]/Rt[-b,2] /;
FreeQ[{a,b},x] && GtQ[a,0] && NegQ[b]
```

2: $\int \frac{1}{\sqrt{a+b\,x^2}}\,dx$ when $a \ngtr 0$

Reference: CRC 278'

Reference: CRC 279'

Derivation: Integration by substitution

Basis: $\frac{1}{\sqrt{a+b\,x^2}} = \text{Subst}\Big[\frac{1}{1-b\,x^2},\ x,\ \frac{x}{\sqrt{a+b\,x^2}}\Big]\ \partial_x \frac{x}{\sqrt{a+b\,x^2}}$

Rule 1.1.3.1.4.1.4.1.2: If $a \ngtr 0$, then

$$\int \frac{1}{\sqrt{a+b x^2}}\,dx \longrightarrow \text{Subst}\left[\int \frac{1}{1-b x^2}\,dx,\ x,\ \frac{x}{\sqrt{a+b x^2}}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^2],x_Symbol] :=
  Subst[Int[1/(1-b*x^2),x],x,x/Sqrt[a+b*x^2]] /;
FreeQ[{a,b},x] && Not[GtQ[a,0]]
```

2. $\int \frac{1}{\sqrt{a+b x^3}}\,dx$

x: $\int \frac{1}{\sqrt{a+b x^3}}\,dx$ when $a>0$

Reference: G&R 3.139

Derivation: Piecewise constant extraction and integration by the Möbius substitution

- **Basis: Let** $q \to \left(\frac{b}{a}\right)^{1/3}$, **then** $\partial_x \frac{\left(1+\sqrt{3}+q x\right)^2 \sqrt{\frac{1+q^3 x^3}{\left(1+\sqrt{3}+q x\right)^4}}}{\sqrt{a+b x^3}} = 0$
- **Basis:** $\frac{1}{\left(1+\sqrt{3}+q x\right)^2 \sqrt{\frac{1+q^3 x^3}{\left(1+\sqrt{3}+q x\right)^4}}} = -\frac{\sqrt{2}\left(1+\sqrt{3}\right)}{3^{1/4} q}\,\text{Subst}\left[\frac{1}{\sqrt{1-x^2}\,\sqrt{1+\left(7+4\sqrt{3}\right)x^2}},\ x,\ \frac{-1+\sqrt{3}-q x}{1+\sqrt{3}+q x}\right]\partial_x \frac{-1+\sqrt{3}-q x}{1+\sqrt{3}+q x}$
- **Note: If** $a>0 \land b>0$, **then** $\text{ArcSin}\left[\frac{-1+\sqrt{3}-\left(\frac{b}{a}\right)^{1/3} x}{1+\sqrt{3}+\left(\frac{b}{a}\right)^{1/3} x}\right]$ **is real when** $\sqrt{a+b x^3}$ **is real.**
- **Note: Although simpler than the following rule, *Mathematica* is unable to validate the result by differentiation.**
- **Rule 1.1.3.1.4.1.4.2.1.1: If** $a>0$, **let** $q \to \left(\frac{b}{a}\right)^{1/3}$, **then**

$$\int \frac{1}{\sqrt{a+b x^3}}\,dx \longrightarrow \frac{\left(1+\sqrt{3}+q x\right)^2 \sqrt{\frac{1+q^3 x^3}{\left(1+\sqrt{3}+q x\right)^4}}}{\sqrt{a+b x^3}} \int \frac{1}{\left(1+\sqrt{3}+q x\right)^2 \sqrt{\frac{1+q^3 x^3}{\left(1+\sqrt{3}+q x\right)^4}}}\,dx$$

$$\to -\frac{\sqrt{2}\left(1+\sqrt{3}\right)\left(1+\sqrt{3}+q\,x\right)^2\sqrt{\frac{1+q^3\,x^3}{\left(1+\sqrt{3}+q\,x\right)^4}}}{3^{1/4}\,q\,\sqrt{a+b\,x^3}}\,\text{Subst}\!\left[\int\frac{1}{\sqrt{1-x^2}\,\sqrt{1+\left(7+4\sqrt{3}\right)x^2}}\,dx,\,x,\,\frac{-1+\sqrt{3}-q\,x}{1+\sqrt{3}+q\,x}\right]$$

$$\to -\frac{\sqrt{2}\left(1+\sqrt{3}\right)\left(1+\sqrt{3}+q\,x\right)^2\sqrt{\frac{1+q^3\,x^3}{\left(1+\sqrt{3}+q\,x\right)^4}}}{3^{1/4}\,q\,\sqrt{a+b\,x^3}}\,\text{EllipticF}\!\left[\text{ArcSin}\!\left[\frac{-1+\sqrt{3}-q\,x}{1+\sqrt{3}+q\,x}\right],\,-7-4\sqrt{3}\right]$$

Program code:

```
(* Int[1/Sqrt[a_+b_.*x_^3],x_Symbol] :=
  With[{q=Rt[b/a,3]},
  -Sqrt[2]*(1+Sqrt[3])*(1+Sqrt[3]+q*x)^2*Sqrt[(1+q^3*x^3)/(1+Sqrt[3]+q*x)^4]/(3^(1/4)*q*Sqrt[a+b*x^3])*
    EllipticF[ArcSin[(-1+Sqrt[3]-q*x)/(1+Sqrt[3]+q*x)],-7-4*Sqrt[3]]] /;
FreeQ[{a,b},x] && PosQ[a] *)
```

1: $\int \frac{1}{\sqrt{a+b\,x^3}}\,dx$ when $a>0$

Reference: G&R 3.139

Derivation: Piecewise constant extraction, integration by the Möbius substitution, and piecewise constant extraction

- **Basis: Let** $q\to\left(\frac{b}{a}\right)^{1/3}$, **then** $\partial_x \frac{\left(1+\sqrt{3}+q\,x\right)^2\sqrt{\frac{1+q^3 x^3}{\left(1+\sqrt{3}+q\,x\right)^4}}}{\sqrt{a+b\,x^3}} == 0$
- **Basis:** $\frac{1}{\left(1+\sqrt{3}+q\,x\right)^2\sqrt{\frac{1+q^3 x^3}{\left(1+\sqrt{3}+q\,x\right)^4}}} == -\frac{2\sqrt{2-\sqrt{3}}}{3^{1/4}\,q}\,\text{Subst}\left[\frac{1}{\sqrt{\left(1-x^2\right)\left(7-4\sqrt{3}+x^2\right)}},\,x,\,\frac{-1+\sqrt{3}-q\,x}{1+\sqrt{3}+q\,x}\right]\partial_x \frac{-1+\sqrt{3}-q\,x}{1+\sqrt{3}+q\,x}$
- **Basis:** $\partial_x \frac{\sqrt{1-x^2}\,\sqrt{7-4\sqrt{3}+x^2}}{\sqrt{\left(1-x^2\right)\left(7-4\sqrt{3}+x^2\right)}} == 0$
- **Note: If** $a>0\land b>0$, **then** $\text{ArcSin}\left[\frac{-1+\sqrt{3}-\left(\frac{b}{a}\right)^{1/3}x}{1+\sqrt{3}+\left(\frac{b}{a}\right)^{1/3}x}\right]$ **is real when** $\sqrt{a+b\,x^3}$ **is real.**
- **Note:** $-7-4\sqrt{3} == -\left(2+\sqrt{3}\right)^2$
- **Warning: The result is discontinuous on the real line when** $x == -\frac{1+\sqrt{3}}{q}$ **where** $q\to\left(\frac{b}{a}\right)^{1/3}$.
- **Rule 1.1.3.1.4.1.4.2.1.1: If** $a>0$, **let** $q\to\frac{r}{s}\to\left(\frac{b}{a}\right)^{1/3}$, **then**

$$\int \frac{1}{\sqrt{a+b\,x^3}}\,dx \longrightarrow \frac{\left(1+\sqrt{3}+q\,x\right)^2\sqrt{\frac{1+q^3 x^3}{\left(1+\sqrt{3}+q\,x\right)^4}}}{\sqrt{a+b\,x^3}}\int \frac{1}{\left(1+\sqrt{3}+q\,x\right)^2\sqrt{\frac{1+q^3 x^3}{\left(1+\sqrt{3}+q\,x\right)^4}}}\,dx$$

$$\longrightarrow -\frac{2\sqrt{2-\sqrt{3}}\,\left(1+\sqrt{3}+q\,x\right)^2\sqrt{\frac{1+q^3 x^3}{\left(1+\sqrt{3}+q\,x\right)^4}}}{3^{1/4}\,q\,\sqrt{a+b\,x^3}}\,\text{Subst}\left[\int \frac{1}{\sqrt{\left(1-x^2\right)\left(7-4\sqrt{3}+x^2\right)}}\,dx,\,x,\,\frac{-1+\sqrt{3}-q\,x}{1+\sqrt{3}+q\,x}\right]$$

$$\to -\frac{2\sqrt{2-\sqrt{3}}\,(1+q\,x)\sqrt{\frac{1-q\,x+q^2\,x^2}{\left(1+\sqrt{3}+q\,x\right)^2}}}{3^{1/4}\,q\,\sqrt{a+b\,x^3}\,\sqrt{\frac{1+q\,x}{\left(1+\sqrt{3}+q\,x\right)^2}}}\,\text{Subst}\left[\int\frac{1}{\sqrt{1-x^2}\,\sqrt{7-4\,\sqrt{3}+x^2}}\,dx,\,x,\,\frac{-1+\sqrt{3}-q\,x}{1+\sqrt{3}+q\,x}\right]$$

$$\to -\frac{2\sqrt{2+\sqrt{3}}\,(1+q\,x)\sqrt{\frac{1-q\,x+q^2\,x^2}{\left(1+\sqrt{3}+q\,x\right)^2}}}{3^{1/4}\,q\,\sqrt{a+b\,x^3}\,\sqrt{\frac{1+q\,x}{\left(1+\sqrt{3}+q\,x\right)^2}}}\,\text{EllipticF}\left[\text{ArcSin}\left[\frac{-1+\sqrt{3}-q\,x}{1+\sqrt{3}+q\,x}\right],\,-7-4\,\sqrt{3}\right]$$

$$\to \frac{2\sqrt{2+\sqrt{3}}\,(s+r\,x)\sqrt{\frac{s^2-r\,s\,x+r^2\,x^2}{\left(\left(1+\sqrt{3}\right)s+r\,x\right)^2}}}{3^{1/4}\,r\,\sqrt{a+b\,x^3}\,\sqrt{\frac{s\,(s+r\,x)}{\left(\left(1+\sqrt{3}\right)s+r\,x\right)^2}}}\,\text{EllipticF}\left[\text{ArcSin}\left[\frac{\left(1-\sqrt{3}\right)s+r\,x}{\left(1+\sqrt{3}\right)s+r\,x}\right],\,-7-4\,\sqrt{3}\right]$$

Program code:

```
(* Int[1/Sqrt[a_+b_.*x_^3],x_Symbol] :=
  With[{q=Rt[a/b,3]},
  2*Sqrt[2+Sqrt[3]]*(q+x)*Sqrt[(q^2-q*x+x^2)/((1+Sqrt[3])*q+x)^2]/
    (3^(1/4)*Sqrt[a+b*x^3]*Sqrt[q*(q+x)/((1+Sqrt[3])*q+x)^2])*
    EllipticF[ArcSin[((1-Sqrt[3])*q+x)/((1+Sqrt[3])*q+x)],-7-4*Sqrt[3]]] /;
FreeQ[{a,b},x] && PosQ[a] && EqQ[b^2,1] *)
```

```
(* Int[1/Sqrt[a_+b_.*x_^3],x_Symbol] :=
  With[{q=Rt[b/a,3]},
  -2*Sqrt[2+Sqrt[3]]*(1+q*x)*Sqrt[(1-q*x+q^2*x^2)/(1+Sqrt[3]+q*x)^2]/
    (3^(1/4)*q*Sqrt[a+b*x^3]*Sqrt[(1+q*x)/(1+Sqrt[3]+q*x)^2])*
    EllipticF[ArcSin[(-1+Sqrt[3]-q*x)/(1+Sqrt[3]+q*x)],-7-4*Sqrt[3]]] /;
FreeQ[{a,b},x] && PosQ[a] *)
```

```
Int[1/Sqrt[a_+b_.*x_^3],x_Symbol] :=
  With[{r=Numer[Rt[b/a,3]], s=Denom[Rt[b/a,3]]},
  2*Sqrt[2+Sqrt[3]]*(s+r*x)*Sqrt[(s^2-r*s*x+r^2*x^2)/((1+Sqrt[3])*s+r*x)^2]/
    (3^(1/4)*r*Sqrt[a+b*x^3]*Sqrt[s*(s+r*x)/((1+Sqrt[3])*s+r*x)^2])*
    EllipticF[ArcSin[((1-Sqrt[3])*s+r*x)/((1+Sqrt[3])*s+r*x)],-7-4*Sqrt[3]]] /;
FreeQ[{a,b},x] && PosQ[a]
```

2: $\int \frac{1}{\sqrt{a+b x^3}} \, dx$ when $a \ngtr 0$

Reference: G&R 3.139

Derivation: Piecewise constant extraction, integration by the Möbius substitution, and piecewise constant extraction

- **Basis: Let** $q \to \left(\frac{b}{a}\right)^{1/3}$**, then** $\partial_x \frac{\left(1-\sqrt{3}+q x\right)^2 \sqrt{-\frac{1+q^3 x^3}{\left(1-\sqrt{3}+q x\right)^4}}}{\sqrt{a+b x^3}} = 0$
- **Basis:** $\frac{1}{\left(1-\sqrt{3}+q x\right)^2 \sqrt{-\frac{1+q^3 x^3}{\left(1-\sqrt{3}+q x\right)^4}}} = \frac{2\sqrt{2-\sqrt{3}}}{3^{1/4} q} \text{Subst}\left[\frac{1}{\sqrt{\left(1-x^2\right)\left(1+\left(7-4\sqrt{3}\right)x^2\right)}}, x, \frac{1+\sqrt{3}+q x}{-1+\sqrt{3}-q x}\right] \partial_x \frac{1+\sqrt{3}+q x}{-1+\sqrt{3}-q x}$
- **Basis:** $\partial_x \frac{\sqrt{1-x^2}\sqrt{1+\left(7-4\sqrt{3}\right)x^2}}{\sqrt{\left(1-x^2\right)\left(1+\left(7-4\sqrt{3}\right)x^2\right)}} = 0$
- **Note: If** $a<0 \land b<0$**, then** $\text{ArcSin}\left[\frac{1+\sqrt{3}+\left(\frac{b}{a}\right)^{1/3} x}{-1+\sqrt{3}-\left(\frac{b}{a}\right)^{1/3} x}\right]$ **is real when** $\sqrt{a+b x^3}$ **is real.**
- **Warning: The result is discontinuous on the real line when** $x = -\frac{1-\sqrt{3}}{q}$ **where** $q \to \left(\frac{b}{a}\right)^{1/3}$**.**
- **Rule 1.1.3.1.4.1.4.2.1: If** $a \ngtr 0$**, let** $q \to \frac{r}{s} \to \left(\frac{b}{a}\right)^{1/3}$**, then**

$$\int \frac{1}{\sqrt{a+b x^3}} \, dx \longrightarrow \frac{\left(1-\sqrt{3}+q x\right)^2 \sqrt{-\frac{1+q^3 x^3}{\left(1-\sqrt{3}+q x\right)^4}}}{\sqrt{a+b x^3}} \int \frac{1}{\left(1-\sqrt{3}+q x\right)^2 \sqrt{-\frac{1+q^3 x^3}{\left(1-\sqrt{3}+q x\right)^4}}} \, dx$$

$$\to \frac{2\sqrt{2-\sqrt{3}}\,\left(1-\sqrt{3}+q\,x\right)^2\sqrt{-\frac{1+q^3\,x^3}{\left(1-\sqrt{3}+q\,x\right)^4}}}{3^{1/4}\,q\,\sqrt{a+b\,x^3}}\,\text{Subst}\left[\int\frac{1}{\sqrt{\left(1-x^2\right)\,\left(1+\left(7-4\,\sqrt{3}\right)\,x^2\right)}}\,dx,\,x,\,\frac{1+\sqrt{3}+q\,x}{-1+\sqrt{3}-q\,x}\right]$$

$$\to -\frac{2\sqrt{2-\sqrt{3}}\,\left(1+q\,x\right)\sqrt{\frac{1-q\,x+q^2\,x^2}{\left(1-\sqrt{3}+q\,x\right)^2}}}{3^{1/4}\,q\,\sqrt{a+b\,x^3}\,\sqrt{-\frac{1+q\,x}{\left(1-\sqrt{3}+q\,x\right)^2}}}\,\text{Subst}\left[\int\frac{1}{\sqrt{1-x^2}\,\sqrt{1+\left(7-4\,\sqrt{3}\right)\,x^2}}\,dx,\,x,\,\frac{1+\sqrt{3}+q\,x}{-1+\sqrt{3}-q\,x}\right]$$

$$\to -\frac{2\sqrt{2-\sqrt{3}}\,\left(1+q\,x\right)\sqrt{\frac{1-q\,x+q^2\,x^2}{\left(1-\sqrt{3}+q\,x\right)^2}}}{3^{1/4}\,q\,\sqrt{a+b\,x^3}\,\sqrt{-\frac{1+q\,x}{\left(1-\sqrt{3}+q\,x\right)^2}}}\,\text{EllipticF}\left[\text{ArcSin}\left[\frac{1+\sqrt{3}+q\,x}{-1+\sqrt{3}-q\,x}\right],\,-7+4\,\sqrt{3}\right]$$

$$\to \frac{2\sqrt{2-\sqrt{3}}\,\left(s+r\,x\right)\sqrt{\frac{s^2-r\,s\,x+r^2\,x^2}{\left(\left(1-\sqrt{3}\right)\,s+r\,x\right)^2}}}{3^{1/4}\,r\,\sqrt{a+b\,x^3}\,\sqrt{-\frac{s\,\left(s+r\,x\right)}{\left(\left(1-\sqrt{3}\right)\,s+r\,x\right)^2}}}\,\text{EllipticF}\left[\text{ArcSin}\left[\frac{\left(1+\sqrt{3}\right)\,s+r\,x}{\left(1-\sqrt{3}\right)\,s+r\,x}\right],\,-7+4\,\sqrt{3}\right]$$

Program code:

```
(* Int[1/Sqrt[a_+b_.*x_^3],x_Symbol] :=
  With[{q=Rt[a/b,3]},
  2*Sqrt[2-Sqrt[3]]*(q+x)*Sqrt[(q^2-q*x+x^2)/((1-Sqrt[3])*q+x)^2]/
    (3^(1/4)*Sqrt[a+b*x^3]*Sqrt[-q*(q+x)/((1-Sqrt[3])*q+x)^2])*
    EllipticF[ArcSin[((1+Sqrt[3])*q+x)/((1-Sqrt[3])*q+x)],-7+4*Sqrt[3]]] /;
FreeQ[{a,b},x] && NegQ[a] && EqQ[b^2,1] *)
```

```
(* Int[1/Sqrt[a_+b_.*x_^3],x_Symbol] :=
  With[{q=Rt[b/a,3]},
  -2*Sqrt[2-Sqrt[3]]*(1+q*x)*Sqrt[(1-q*x+q^2*x^2)/(1-Sqrt[3]+q*x)^2]/
    (3^(1/4)*q*Sqrt[a+b*x^3]*Sqrt[-(1+q*x)/(1-Sqrt[3]+q*x)^2])*
    EllipticF[ArcSin[(1+Sqrt[3]+q*x)/(-1+Sqrt[3]-q*x)],-7+4*Sqrt[3]]] /;
FreeQ[{a,b},x] && NegQ[a] *)
```

```
Int[1/Sqrt[a_+b_.*x_^3],x_Symbol] :=
  With[{r=Numer[Rt[b/a,3]], s=Denom[Rt[b/a,3]]},
  2*Sqrt[2-Sqrt[3]]*(s+r*x)*Sqrt[(s^2-r*s*x+r^2*x^2)/((1-Sqrt[3])*s+r*x)^2]/
    (3^(1/4)*r*Sqrt[a+b*x^3]*Sqrt[-s*(s+r*x)/((1-Sqrt[3])*s+r*x)^2])*
    EllipticF[ArcSin[((1+Sqrt[3])*s+r*x)/((1-Sqrt[3])*s+r*x)],-7+4*Sqrt[3]]] /;
FreeQ[{a,b},x] && NegQ[a]
```

3. $\int \frac{1}{\sqrt{a+b\,x^4}}\,dx$

1: $\int \frac{1}{\sqrt{a+b\,x^4}}\,dx$ when $\frac{b}{a}>0$

- **Reference: G&R 3.166.1**

- **Derivation: Piecewise constant extraction**

- **Basis:** $\partial_x \frac{\left(1+q^2 x^2\right)\sqrt{\frac{a+b\,x^4}{a\left(1+q^2 x^2\right)^2}}}{\sqrt{a+b\,x^4}} == 0$

- **Contributed by Martin Welz on 12 August 2016**

- **Rule 1.1.3.1.4.1.4.3.1: If** $\frac{b}{a}>0$**, let** $q\to\left(\frac{b}{a}\right)^{1/4}$**, then**

$$\int \frac{1}{\sqrt{a+b\,x^4}}\,dx \longrightarrow \frac{\left(1+q^2 x^2\right)\sqrt{\frac{a+b\,x^4}{a\left(1+q^2 x^2\right)^2}}}{\sqrt{a+b\,x^4}}\int \frac{1}{\left(1+q^2 x^2\right)\sqrt{\frac{a+b\,x^4}{a\left(1+q^2 x^2\right)^2}}}\,dx$$

$$\longrightarrow \frac{\left(1+q^2 x^2\right)\sqrt{\frac{a+b\,x^4}{a\left(1+q^2 x^2\right)^2}}}{2\,q\,\sqrt{a+b\,x^4}}\,\text{EllipticF}\left[2\,\text{ArcTan}[q\,x],\ \frac{1}{2}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^4],x_Symbol] :=
  With[{q=Rt[b/a,4]},
  (1+q^2*x^2)*Sqrt[(a+b*x^4)/(a*(1+q^2*x^2)^2)]/(2*q*Sqrt[a+b*x^4])*EllipticF[2*ArcTan[q*x],1/2]] /;
FreeQ[{a,b},x] && PosQ[b/a]
```

2. $\int \frac{1}{\sqrt{a+b\,x^4}}\,dx$ when $\frac{b}{a}\not>0$

1: $\int \frac{1}{\sqrt{a+b\,x^4}}\,dx$ when $\frac{b}{a}\not>0\ \bigwedge\ a>0$

- **Rule 1.1.3.1.4.1.4.3.2.1: If** $\frac{b}{a}\not>0\ \bigwedge\ a>0$**, then**

$$\int \frac{1}{\sqrt{a+b\,x^4}}\,dx \to \frac{1}{a^{1/4}\,(-b)^{1/4}}\,\text{EllipticF}\left[\text{ArcSin}\left[\frac{(-b)^{1/4}\,x}{a^{1/4}}\right],\,-1\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^4],x_Symbol] :=
  EllipticF[ArcSin[Rt[-b,4]*x/Rt[a,4]],-1]/(Rt[a,4]*Rt[-b,4]) /;
FreeQ[{a,b},x] && NegQ[b/a] && GtQ[a,0]
```

2: $\int \frac{1}{\sqrt{a+b\,x^4}}\,dx$ when $a<0\,\wedge\,b>0$

Reference: G&R 3.152.3+

Note: Not sure if the shorter rule is valid for all q.

Rule 1.1.3.1.4.1.4.3.2.2: If $a<0\,\wedge\,b>0$, let $q\to\sqrt{-a\,b}$, then

$$\int \frac{1}{\sqrt{a+b\,x^4}}\,dx \to \frac{\sqrt{\frac{a-q\,x^2}{a+q\,x^2}}\,\sqrt{\frac{a+q\,x^2}{q}}}{\sqrt{2}\,\sqrt{a+b\,x^4}\,\sqrt{\frac{a}{a+q\,x^2}}}\,\text{EllipticF}\left[\text{ArcSin}\left[\frac{x}{\sqrt{\frac{a+q\,x^2}{2\,q}}}\right],\,\frac{1}{2}\right]$$

$$\int \frac{1}{\sqrt{a+b\,x^4}}\,dx \to \frac{\sqrt{-a+q\,x^2}\,\sqrt{\frac{a+q\,x^2}{q}}}{\sqrt{2}\,\sqrt{-a}\,\sqrt{a+b\,x^4}}\,\text{EllipticF}\left[\text{ArcSin}\left[\frac{x}{\sqrt{\frac{a+q\,x^2}{2\,q}}}\right],\,\frac{1}{2}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^4],x_Symbol] :=
  With[{q=Rt[-a*b,2]},
  Sqrt[-a+q*x^2]*Sqrt[(a+q*x^2)/q]/(Sqrt[2]*Sqrt[-a]*Sqrt[a+b*x^4])*
    EllipticF[ArcSin[x/Sqrt[(a+q*x^2)/(2*q)]],1/2] /;
  IntegerQ[q]] /;
FreeQ[{a,b},x] && LtQ[a,0] && GtQ[b,0]
```

```
Int[1/Sqrt[a_+b_.*x_^4],x_Symbol] :=
  With[{q=Rt[-a*b,2]},
  Sqrt[(a-q*x^2)/(a+q*x^2)]*Sqrt[(a+q*x^2)/q]/(Sqrt[2]*Sqrt[a+b*x^4]*Sqrt[a/(a+q*x^2)])*
    EllipticF[ArcSin[x/Sqrt[(a+q*x^2)/(2*q)]],1/2]] /;
FreeQ[{a,b},x] && LtQ[a,0] && GtQ[b,0]
```

3: $\int \frac{1}{\sqrt{a+b x^4}} \, dx$ when $\frac{b}{a} \not> 0 \bigwedge a \not> 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{1+\frac{b x^4}{a}}}{\sqrt{a+b x^4}} == 0$
- **Rule 1.1.3.1.4.1.4.3.2.3: If** $\frac{b}{a} \not> 0 \bigwedge a \not> 0$**, then**

$$\int \frac{1}{\sqrt{a+b x^4}} \, dx \longrightarrow \frac{\sqrt{1+\frac{b x^4}{a}}}{\sqrt{a+b x^4}} \int \frac{1}{\sqrt{1+\frac{b x^4}{a}}} \, dx$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^4],x_Symbol] :=
  Sqrt[1+b*x^4/a]/Sqrt[a+b*x^4]*Int[1/Sqrt[1+b*x^4/a],x] /;
FreeQ[{a,b},x] && NegQ[b/a] && Not[GtQ[a,0]]
```

4: $\int \frac{1}{\sqrt{a+b x^6}}\,dx$

Derivation: Piecewise constant extraction and integration by the substitution

- **Basis: Let** $q \to \left(\frac{b}{a}\right)^{1/3}$, **then** $\partial_x \frac{x\,(1+q x^2)\sqrt{\frac{1-q x^2+q^2 x^4}{\left(1+\left(1+\sqrt{3}\right) q x^2\right)^2}}}{\sqrt{a+b x^6}\,\sqrt{\frac{q x^2\,(1+q x^2)}{\left(1+\left(1+\sqrt{3}\right) q x^2\right)^2}}} == 0$

- **Basis:** $\frac{\sqrt{\frac{q x^2\,(1+q x^2)}{\left(1+\left(1+\sqrt{3}\right) q x^2\right)^2}}}{x\,(1+q x^2)\sqrt{\frac{1-q x^2+q^2 x^4}{\left(1+\left(1+\sqrt{3}\right) q x^2\right)^2}}} == -\frac{1}{3^{1/4}}\,\text{Subst}\left[\frac{1}{\sqrt{1-x^2}\,\sqrt{2-\sqrt{3}+\left(2+\sqrt{3}\right) x^2}},\,x,\,\frac{1+\left(1-\sqrt{3}\right) q x^2}{1+\left(1+\sqrt{3}\right) q x^2}\right]\partial_x \frac{1+\left(1-\sqrt{3}\right) q x^2}{1+\left(1+\sqrt{3}\right) q x^2}$

- **Rule 1.1.3.1.4.1.4.4: Let** $q \to \frac{r}{s} \to \left(\frac{b}{a}\right)^{1/3}$, **then**

$$\int \frac{1}{\sqrt{a+b x^6}}\,dx \to \frac{x\left(1+q x^2\right)\sqrt{\frac{1-q x^2+q^2 x^4}{\left(1+\left(1+\sqrt{3}\right) q x^2\right)^2}}}{\sqrt{a+b x^6}\,\sqrt{\frac{q x^2\,(1+q x^2)}{\left(1+\left(1+\sqrt{3}\right) q x^2\right)^2}}}\int \frac{\sqrt{\frac{q x^2\,(1+q x^2)}{\left(1+\left(1+\sqrt{3}\right) q x^2\right)^2}}}{x\left(1+q x^2\right)\sqrt{\frac{1-q x^2+q^2 x^4}{\left(1+\left(1+\sqrt{3}\right) q x^2\right)^2}}}\,dx$$

$$\to -\frac{x\left(1+q x^2\right)\sqrt{\frac{1-q x^2+q^2 x^4}{\left(1+\left(1+\sqrt{3}\right) q x^2\right)^2}}}{3^{1/4}\sqrt{a+b x^6}\,\sqrt{\frac{q x^2\,(1+q x^2)}{\left(1+\left(1+\sqrt{3}\right) q x^2\right)^2}}}\,\text{Subst}\left[\int \frac{1}{\sqrt{1-x^2}\,\sqrt{2-\sqrt{3}+\left(2+\sqrt{3}\right) x^2}}\,dx,\,x,\,\frac{1+\left(1-\sqrt{3}\right) q x^2}{1+\left(1+\sqrt{3}\right) q x^2}\right]$$

$$\to \frac{x\left(1+q x^2\right)\sqrt{\frac{1-q x^2+q^2 x^4}{\left(1+\left(1+\sqrt{3}\right) q x^2\right)^2}}}{2\times 3^{1/4}\sqrt{a+b x^6}\,\sqrt{\frac{q x^2\,(1+q x^2)}{\left(1+\left(1+\sqrt{3}\right) q x^2\right)^2}}}\,\text{EllipticF}\left[\text{ArcCos}\left[\frac{1+\left(1-\sqrt{3}\right) q x^2}{1+\left(1+\sqrt{3}\right) q x^2}\right],\,\frac{2+\sqrt{3}}{4}\right]$$

$$\rightarrow \frac{x\left(s+r x^2\right)\sqrt{\frac{s^2-r s x^2+r^2 x^4}{\left(s+\left(1+\sqrt{3}\right) r x^2\right)^2}}}{2\times 3^{1/4} s\sqrt{a+b x^6}\sqrt{\frac{r x^2\left(s+r x^2\right)}{\left(s+\left(1+\sqrt{3}\right) r x^2\right)^2}}}\text{EllipticF}\left[\text{ArcCos}\left[\frac{s+\left(1-\sqrt{3}\right) r x^2}{s+\left(1+\sqrt{3}\right) r x^2}\right],\frac{2+\sqrt{3}}{4}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^6],x_Symbol] :=
  With[{r=Numer[Rt[b/a,3]], s=Denom[Rt[b/a,3]]},
  x*(s+r*x^2)*Sqrt[(s^2-r*s*x^2+r^2*x^4)/(s+(1+Sqrt[3])*r*x^2)^2]/
    (2*3^(1/4)*s*Sqrt[a+b*x^6]*Sqrt[r*x^2*(s+r*x^2)/(s+(1+Sqrt[3])*r*x^2)^2])*
    EllipticF[ArcCos[(s+(1-Sqrt[3])*r*x^2)/(s+(1+Sqrt[3])*r*x^2)],(2+Sqrt[3])/4]] /;
FreeQ[{a,b},x]
```

5: $\int \frac{1}{\sqrt{a+b x^8}}\,dx$

Derivation: Algebraic expansion

- **Basis:** $\frac{1}{\sqrt{a+b x^8}} == \frac{1-\left(\frac{b}{a}\right)^{1/4} x^2}{2\sqrt{a+b x^8}}+\frac{1+\left(\frac{b}{a}\right)^{1/4} x^2}{2\sqrt{a+b x^8}}$
- **Note: Integrands are of the form $\frac{c+d x^2}{\sqrt{a+b x^8}}$ where $b c^4-a d^4 == 0$ for which there is a terminal rule.**

Rule 1.1.3.1.4.1.4.5:

$$\int \frac{1}{\sqrt{a+b x^8}}\,dx \rightarrow \frac{1}{2}\int \frac{1-\left(\frac{b}{a}\right)^{1/4} x^2}{\sqrt{a+b x^8}}\,dx+\frac{1}{2}\int \frac{1+\left(\frac{b}{a}\right)^{1/4} x^2}{\sqrt{a+b x^8}}\,dx$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^8],x_Symbol] :=
  1/2*Int[(1-Rt[b/a,4]*x^2)/Sqrt[a+b*x^8],x] +
  1/2*Int[(1+Rt[b/a,4]*x^2)/Sqrt[a+b*x^8],x] /;
FreeQ[{a,b},x]
```

5. $\int \frac{1}{\left(a+b x^2\right)^{1/4}} \, dx$

1. $\int \frac{1}{\left(a+b x^2\right)^{1/4}} \, dx$ when $a \not< 0$

1. $\int \frac{1}{\left(a+b x^2\right)^{1/4}} \, dx$ when $a > 0$

1: $\int \frac{1}{\left(a+b x^2\right)^{1/4}} \, dx$ when $a > 0 \bigwedge \frac{b}{a} > 0$

Reference: G&R 2.110.1, CRC 88b

Derivation: Binomial recurrence 1b

- **Rule 1.1.3.1.4.1.5.1.1.1: If $a > 0 \bigwedge \frac{b}{a} > 0$, then**

$$\int \frac{1}{\left(a+b x^2\right)^{1/4}} \, dx \longrightarrow \frac{2 x}{\left(a+b x^2\right)^{1/4}} - a \int \frac{1}{\left(a+b x^2\right)^{5/4}} \, dx$$

Program code:

```
Int[1/(a_+b_.*x_^2)^(1/4),x_Symbol] :=
  2*x/(a+b*x^2)^(1/4) - a*Int[1/(a+b*x^2)^(5/4),x] /;
FreeQ[{a,b},x] && GtQ[a,0] && PosQ[b/a]
```

2: $\int \frac{1}{(a+b x^2)^{1/4}} dx$ when $a>0 \bigwedge \frac{b}{a} \not> 0$

- **Rule 1.1.3.1.4.1.5.1.1.2: If $a>0 \bigwedge \frac{b}{a} \not> 0$, then**

$$\int \frac{1}{(a+b x^2)^{1/4}} dx \longrightarrow \frac{2}{a^{1/4} \sqrt{-\frac{b}{a}}} \text{EllipticE}\left[\frac{1}{2} \text{ArcSin}\left[\sqrt{-\frac{b}{a}}\, x\right], 2\right]$$

Program code:

```
Int[1/(a_+b_.*x_^2)^(1/4),x_Symbol] :=
  2/(a^(1/4)*Rt[-b/a,2])*EllipticE[1/2*ArcSin[Rt[-b/a,2]*x],2] /;
FreeQ[{a,b},x] && GtQ[a,0] && NegQ[b/a]
```

2: $\int \frac{1}{(a+b x^2)^{1/4}} dx$ when $a \not> 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\left(1+\frac{b x^2}{a}\right)^{1/4}}{(a+b x^2)^{1/4}} == 0$

Rule 1.1.3.1.4.1.5.1.2: If $a \not> 0$, then

$$\int \frac{1}{(a+b x^2)^{1/4}} dx \longrightarrow \frac{\left(1+\frac{b x^2}{a}\right)^{1/4}}{(a+b x^2)^{1/4}} \int \frac{1}{\left(1+\frac{b x^2}{a}\right)^{1/4}} dx$$

Program code:

```
Int[1/(a_+b_.*x_^2)^(1/4),x_Symbol] :=
  (1+b*x^2/a)^(1/4)/(a+b*x^2)^(1/4)*Int[1/(1+b*x^2/a)^(1/4),x] /;
FreeQ[{a,b},x] && PosQ[a]
```

2: $\int \frac{1}{(a+b x^2)^{1/4}} \, dx$ when $a \ngtr 0$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x \frac{\sqrt{-\frac{b x^2}{a}}}{x} = 0$
- **Basis:** $\frac{x}{\sqrt{-\frac{b x^2}{a}} \, (a+b x^2)^{1/4}} = \frac{2}{b} \text{Subst}\left[\frac{x^2}{\sqrt{1-\frac{x^4}{a}}}, x, (a+b x^2)^{1/4}\right] \partial_x (a+b x^2)^{1/4}$

Rule 1.1.3.1.4.1.5.2: If $a \ngtr 0$, then

$$\int \frac{1}{(a+b x^2)^{1/4}} \, dx \to \frac{\sqrt{-\frac{b x^2}{a}}}{x} \int \frac{x}{\sqrt{-\frac{b x^2}{a}} \, (a+b x^2)^{1/4}} \, dx \to \frac{2\sqrt{-\frac{b x^2}{a}}}{b x} \text{Subst}\left[\int \frac{x^2}{\sqrt{1-\frac{x^4}{a}}} \, dx, x, (a+b x^2)^{1/4}\right]$$

Program code:

```
Int[1/(a_+b_.*x_^2)^(1/4),x_Symbol] :=
  2*Sqrt[-b*x^2/a]/(b*x)*Subst[Int[x^2/Sqrt[1-x^4/a],x],x,(a+b*x^2)^(1/4)] /;
FreeQ[{a,b},x] && NegQ[a]
```

6. $\int \frac{1}{(a+b x^2)^{3/4}} \, dx$

1. $\int \frac{1}{(a+b x^2)^{3/4}} \, dx$ when $a \nless 0$

1. $\int \frac{1}{(a+b x^2)^{3/4}} \, dx$ when $a > 0$

1: $\int \frac{1}{(a+b x^2)^{3/4}} \, dx$ when $a > 0 \bigwedge \frac{b}{a} > 0$

Contributed by Martin Welz on 7 August 2016

- **Rule 1.1.3.1.4.1.6.1.1.1: If $a > 0 \bigwedge \frac{b}{a} > 0$, then**

$$\int \frac{1}{\left(a+b\,x^2\right)^{3/4}}\,dx \longrightarrow \frac{2}{a^{3/4}\sqrt{\frac{b}{a}}}\,\text{EllipticF}\left[\frac{1}{2}\,\text{ArcTan}\left[\sqrt{\frac{b}{a}}\,x\right],\,2\right]$$

Program code:

```
Int[1/(a_+b_.*x_^2)^(3/4),x_Symbol] :=
  2/(a^(3/4)*Rt[b/a,2])*EllipticF[1/2*ArcTan[Rt[b/a,2]*x],2] /;
FreeQ[{a,b},x] && GtQ[a,0] && PosQ[b/a]
```

2: $\int \frac{1}{\left(a+b\,x^2\right)^{3/4}}\,dx$ when $a>0 \bigwedge \frac{b}{a} \not> 0$

$$\int \frac{1}{\left(a+b\,x^2\right)^{3/4}}\,dx \longrightarrow \frac{2}{a^{3/4}\sqrt{-\frac{b}{a}}}\,\text{EllipticF}\left[\frac{1}{2}\,\text{ArcSin}\left[\sqrt{-\frac{b}{a}}\,x\right],\,2\right]$$

- **Rule 1.1.3.1.4.1.6.1.1.2: If** $a>0 \bigwedge \frac{b}{a} \not> 0$**, then**

Program code:

```
Int[1/(a_+b_.*x_^2)^(3/4),x_Symbol] :=
  2/(a^(3/4)*Rt[-b/a,2])*EllipticF[1/2*ArcSin[Rt[-b/a,2]*x],2] /;
FreeQ[{a,b},x] && GtQ[a,0] && NegQ[b/a]
```

2: $\int \frac{1}{\left(a+b\,x^2\right)^{3/4}}\,dx$ when $a \not< 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\left(1+\frac{b\,x^2}{a}\right)^{3/4}}{\left(a+b\,x^2\right)^{3/4}} == 0$

Rule 1.1.3.1.4.1.6.1.2: If $a \not< 0$**, then**

$$\int \frac{1}{\left(a+b\,x^2\right)^{3/4}}\,dx \longrightarrow \frac{\left(1+\frac{b\,x^2}{a}\right)^{3/4}}{\left(a+b\,x^2\right)^{3/4}} \int \frac{1}{\left(1+\frac{b\,x^2}{a}\right)^{3/4}}\,dx$$

Program code:

```
Int[1/(a_+b_.*x_^2)^(3/4),x_Symbol] :=
  (1+b*x^2/a)^(3/4)/(a+b*x^2)^(3/4)*Int[1/(1+b*x^2/a)^(3/4),x] /;
FreeQ[{a,b},x] && PosQ[a]
```

2: $\int \frac{1}{\left(a+b\,x^2\right)^{3/4}}\,dx$ when $a<0$

Derivation: Piecewise constant extranction and integration by substitution

- Basis: $\partial_x \frac{\sqrt{-\frac{b\,x^2}{a}}}{x} == 0$
- Basis: $\frac{x}{\sqrt{-\frac{b\,x^2}{a}}\,\left(a+b\,x^2\right)^{3/4}} == \frac{2}{b}\,\text{Subst}\left[\frac{1}{\sqrt{1-\frac{x^4}{a}}},\,x,\,\left(a+b\,x^2\right)^{1/4}\right]\,\partial_x\left(a+b\,x^2\right)^{1/4}$

Rule 1.1.3.1.4.1.6.2: If $a<0$, then

$$\int \frac{1}{\left(a+b\,x^2\right)^{3/4}}\,dx \longrightarrow \frac{\sqrt{-\frac{b\,x^2}{a}}}{x}\int \frac{x}{\sqrt{-\frac{b\,x^2}{a}}\,\left(a+b\,x^2\right)^{3/4}}\,dx \longrightarrow \frac{2\sqrt{-\frac{b\,x^2}{a}}}{b\,x}\,\text{Subst}\left[\int \frac{1}{\sqrt{1-\frac{x^4}{a}}}\,dx,\,x,\,\left(a+b\,x^2\right)^{1/4}\right]$$

Program code:

```
Int[1/(a_+b_.*x_^2)^(3/4),x_Symbol] :=
  2*Sqrt[-b*x^2/a]/(b*x)*Subst[Int[1/Sqrt[1-x^4/a],x],x,(a+b*x^2)^(1/4)] /;
FreeQ[{a,b},x] && NegQ[a]
```

7: $\int \frac{1}{\left(a+b\,x^2\right)^{1/3}}\,dx$

Derivation: Integration by substitution and piecewise constant extraction

- **Basis:** $\frac{1}{\left(a+b\,x^2\right)^{1/3}} == \frac{3\sqrt{b\,x^2}}{2\,b\,x}\,\text{Subst}\left[\frac{x}{\sqrt{-a+x^3}},\ x,\ \left(a+b\,x^2\right)^{1/3}\right]\,\partial_x\left(a+b\,x^2\right)^{1/3}$
- **Basis:** $\partial_x\frac{\sqrt{b\,x^2}}{x} == 0$

Rule 1.1.3.1.4.1.7:

$$\int \frac{1}{\left(a+b\,x^2\right)^{1/3}}\,dx \longrightarrow \frac{3\sqrt{b\,x^2}}{2\,b\,x}\,\text{Subst}\left[\int \frac{x}{\sqrt{-a+x^3}}\,dx,\ x,\ \left(a+b\,x^2\right)^{1/3}\right]$$

Program code:

```
Int[1/(a_+b_.*x_^2)^(1/3),x_Symbol] :=
  3*Sqrt[b*x^2]/(2*b*x)*Subst[Int[x/Sqrt[-a+x^3],x],x,(a+b*x^2)^(1/3)] /;
FreeQ[{a,b},x]
```

8: $\int \frac{1}{\left(a+b\,x^2\right)^{2/3}}\,dx$

Derivation: Integration by substitution and piecewise constant extraction

- **Basis:** $\frac{1}{\left(a+b\,x^2\right)^{2/3}} == \frac{3\sqrt{b\,x^2}}{2\,b\,x}\,\text{Subst}\left[\frac{1}{\sqrt{-a+x^3}},\ x,\ \left(a+b\,x^2\right)^{1/3}\right]\,\partial_x\left(a+b\,x^2\right)^{1/3}$
- **Basis:** $\partial_x\frac{\sqrt{b\,x^2}}{x} == 0$

Rule 1.1.3.1.4.1.8:

$$\int \frac{1}{\left(a+b\,x^2\right)^{2/3}}\,dx \longrightarrow \frac{3\sqrt{b\,x^2}}{2\,b\,x}\,\text{Subst}\left[\int \frac{1}{\sqrt{-a+x^3}}\,dx,\ x,\ \left(a+b\,x^2\right)^{1/3}\right]$$

Program code:

```
Int[1/(a_+b_.*x_^2)^(2/3),x_Symbol] :=
  3*Sqrt[b*x^2]/(2*b*x)*Subst[Int[1/Sqrt[-a+x^3],x],x,(a+b*x^2)^(1/3)] /;
FreeQ[{a,b},x]
```

9: $\int \frac{1}{(a+b x^4)^{3/4}} \, dx$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{x^3 \left(1+\frac{a}{b x^4}\right)^{3/4}}{(a+b x^4)^{3/4}} == 0$

Rule 1.1.3.1.4.1.9:

$$\int \frac{1}{(a+b x^4)^{3/4}} \, dx \longrightarrow \frac{x^3 \left(1+\frac{a}{b x^4}\right)^{3/4}}{(a+b x^4)^{3/4}} \int \frac{1}{x^3 \left(1+\frac{a}{b x^4}\right)^{3/4}} \, dx$$

Program code:

```
Int[1/(a_+b_.*x_^4)^(3/4),x_Symbol] :=
  x^3*(1+a/(b*x^4))^(3/4)/(a+b*x^4)^(3/4)*Int[1/(x^3*(1+a/(b*x^4))^(3/4)),x] /;
FreeQ[{a,b},x]
```

10: $\int \frac{1}{(a+b x^2)^{1/6}} \, dx$

Derivation: Binomial recurrence 2b

Rule 1.1.3.1.4.1.10:

$$\int \frac{1}{(a+b x^2)^{1/6}} \, dx \longrightarrow \frac{3 x}{2 (a+b x^2)^{1/6}} - \frac{a}{2} \int \frac{1}{(a+b x^2)^{7/6}} \, dx$$

Program code:

```
Int[1/(a_+b_.*x_^2)^(1/6),x_Symbol] :=
  3*x/(2*(a+b*x^2)^(1/6)) - a/2*Int[1/(a+b*x^2)^(7/6),x] /;
FreeQ[{a,b},x]
```

11: $\int \frac{1}{(a+b x^3)^{1/3}} \, dx$

Rule 1.1.3.1.4.1.11:

$$\int \frac{1}{\left(a+b x^3\right)^{1/3}} dx \longrightarrow \frac{\operatorname{ArcTan}\left[\frac{1+\frac{2 b^{1/3} x}{\left(a+b x^3\right)^{1/3}}}{\sqrt{3}}\right]}{\sqrt{3}\, b^{1/3}}-\frac{\operatorname{Log}\left[\left(a+b x^3\right)^{1/3}-b^{1/3} x\right]}{2 b^{1/3}}$$

Program code:

```
Int[1/(a_+b_.*x_^3)^(1/3),x_Symbol] :=
  ArcTan[(1+2*Rt[b,3]*x/(a+b*x^3)^(1/3))/Sqrt[3]]/(Sqrt[3]*Rt[b,3]) - Log[(a+b*x^3)^(1/3)-Rt[b,3]*x]/(2*Rt[b,3]) /;
FreeQ[{a,b},x]
```

12. $\int (a+b x^n)^p \, dx$ when $n \in \mathbb{Z}^+ \bigwedge -1<p<0 \bigwedge p \neq -\frac{1}{2} \bigwedge \text{Denominator}\left[p+\frac{1}{n}\right] < \text{Denominator}[p]$

1: $\int (a+b x^n)^p \, dx$ when $n \in \mathbb{Z}^+ \bigwedge -1<p<0 \bigwedge p \neq -\frac{1}{2} \bigwedge p+\frac{1}{n} \in \mathbb{Z}$

Derivation: Integration by substitution

- **Basis: If** $n \in \mathbb{Z}^+ \bigwedge p+\frac{1}{n} \in \mathbb{Z}$, **then** $(a+b x^n)^p == a^{p+\frac{1}{n}} \text{Subst}\left[\frac{1}{(1-b x^n)^{p+\frac{1}{n}+1}}, x, \frac{x}{(a+b x^n)^{1/n}}\right] \partial_x \frac{x}{(a+b x^n)^{1/n}}$

- **Rule 1.1.3.1.4.1.12.1: If** $n \in \mathbb{Z}^+ \bigwedge -1<p<0 \bigwedge p \neq -\frac{1}{2} \bigwedge p+\frac{1}{n} \in \mathbb{Z}$, **then**

$$\int (a+b x^n)^p \, dx \longrightarrow a^{p+\frac{1}{n}} \text{Subst}\left[\int \frac{1}{(1-b x^n)^{p+\frac{1}{n}+1}} dx, x, \frac{x}{(a+b x^n)^{1/n}}\right]$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  a^(p+1/n)*Subst[Int[1/(1-b*x^n)^(p+1/n+1),x],x,x/(a+b*x^n)^(1/n)] /;
FreeQ[{a,b},x] && IGtQ[n,0] && LtQ[-1,p,0] && NeQ[p,-1/2] && IntegerQ[p+1/n]
```

2: $\int (a+b\,x^n)^p\,dx$ when $n\in\mathbb{Z}^+ \bigwedge -1<p<0 \bigwedge p\neq -\frac{1}{2} \bigwedge \text{Denominator}\left[p+\frac{1}{n}\right]<\text{Denominator}[p]$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x\left(\left(\frac{a}{a+b\,x^n}\right)^{p+\frac{1}{n}}(a+b\,x^n)^{p+\frac{1}{n}}\right)=0$
- **Basis: If** $n\in\mathbb{Z}$, **then** $\frac{1}{\left(\frac{a}{a+b\,x^n}\right)^{p+\frac{1}{n}}(a+b\,x^n)^{\frac{1}{n}}}=\text{Subst}\left[\frac{1}{(1-b\,x^n)^{p+\frac{1}{n}+1}},\ x,\ \frac{x}{(a+b\,x^n)^{1/n}}\right]\partial_x\frac{x}{(a+b\,x^n)^{1/n}}$
- **Rule 1.1.3.1.4.1.12.2: If** $n\in\mathbb{Z}^+ \bigwedge -1<p<0 \bigwedge p\neq -\frac{1}{2} \bigwedge \text{Denominator}\left[p+\frac{1}{n}\right]<\text{Denominator}[p]$, **then**

$$\int (a+b\,x^n)^p\,dx \longrightarrow \left(\frac{a}{a+b\,x^n}\right)^{p+\frac{1}{n}}(a+b\,x^n)^{p+\frac{1}{n}}\int\frac{1}{\left(\frac{a}{a+b\,x^n}\right)^{p+\frac{1}{n}}(a+b\,x^n)^{\frac{1}{n}}}\,dx$$

$$\longrightarrow \left(\frac{a}{a+b\,x^n}\right)^{p+\frac{1}{n}}(a+b\,x^n)^{p+\frac{1}{n}}\,\text{Subst}\left[\int\frac{1}{(1-b\,x^n)^{p+\frac{1}{n}+1}}\,dx,\ x,\ \frac{x}{(a+b\,x^n)^{1/n}}\right]$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  (a/(a+b*x^n))^(p+1/n)*(a+b*x^n)^(p+1/n)*Subst[Int[1/(1-b*x^n)^(p+1/n+1),x],x,x/(a+b*x^n)^(1/n)] /;
FreeQ[{a,b},x] && IGtQ[n,0] && LtQ[-1,p,0] && NeQ[p,-1/2] && LtQ[Denominator[p+1/n],Denominator[p]]
```

2: $\int (a+b\,x^n)^p\,dx$ when $n\in\mathbb{Z}^-$

Derivation: Integration by substitution

- **Basis: If** $n\in\mathbb{Z}$, **then** $F[x^n]=-\text{Subst}\left[\frac{F[x^{-n}]}{x^2},\ x,\ \frac{1}{x}\right]\partial_x\frac{1}{x}$
- **Rule 1.1.3.1.4.2: If** $n\in\mathbb{Z}^-$, **then**

$$\int (a+b\,x^n)^p\,dx \longrightarrow -\text{Subst}\left[\int\frac{(a+b\,x^{-n})^p}{x^2}\,dx,\ x,\ \frac{1}{x}\right]$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  -Subst[Int[(a+b*x^(-n))^p/x^2,x],x,1/x] /;
FreeQ[{a,b,p},x] && ILtQ[n,0]
```

5: $\int (a + b\,x^n)^p\,dx$ when $n \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$**, then** $F[x^n] = k\,\text{Subst}\left[x^{k-1}\,F\left[x^{k\,n}\right],\ x,\ x^{1/k}\right]\,\partial_x x^{1/k}$

Rule 1.1.3.1.5: If $n \in \mathbb{F}$**, let** $k \to \text{Denominator}[n]$**, then**

$$\int (a + b\,x^n)^p\,dx \;\longrightarrow\; k\,\text{Subst}\left[\int x^{k-1}\left(a + b\,x^{k\,n}\right)^p dx,\ x,\ x^{1/k}\right]$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k-1)*(a+b*x^(k*n))^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,p},x] && FractionQ[n]
```

6: $\int (a + b\,x^n)^p\,dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.3.1.6: If $p \in \mathbb{Z}^+$**, then**

$$\int (a + b\,x^n)^p\,dx \;\longrightarrow\; \int \text{ExpandIntegrand}[(a + b\,x^n)^p,\ x]\,dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,n},x] && IGtQ[p,0]
```

H. $\int (a + b\,x^n)^p\,dx$ when $p \notin \mathbb{Z}^+ \bigwedge \frac{1}{n} \notin \mathbb{Z} \bigwedge \frac{1}{n} + p \notin \mathbb{Z}^-$

1: $\int (a + b\,x^n)^p\,dx$ when $p \notin \mathbb{Z}^+ \bigwedge \frac{1}{n} \notin \mathbb{Z} \bigwedge \frac{1}{n} + p \notin \mathbb{Z}^- \bigwedge (p \in \mathbb{Z}^- \bigvee a > 0)$

Note: If $t = r + 1 \land r \in \mathbb{Z}$**, then** $\text{Hypergeometric2F1}[r,\ s,\ t,\ z] = \text{Hypergeometric2F1}[s,\ r,\ t,\ z]$ **are elementary or undefined.**

Rule 1.1.3.1.7.1: If $p \notin \mathbb{Z}^+ \bigwedge \frac{1}{n} \notin \mathbb{Z} \bigwedge \frac{1}{n} + p \notin \mathbb{Z}^- \bigwedge (p \in \mathbb{Z}^- \bigvee a > 0)$**, then**

$$\int (a+b\,x^n)^p\,dx \longrightarrow a^p\,x\,\text{Hypergeometric2F1}\left[-p,\,\frac{1}{n},\,\frac{1}{n}+1,\,-\frac{b\,x^n}{a}\right]$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  a^p*x*Hypergeometric2F1[-p,1/n,1/n+1,-b*x^n/a] /;
FreeQ[{a,b,n,p},x] && Not[IGtQ[p,0]] && Not[IntegerQ[1/n]] && Not[ILtQ[Simplify[1/n+p],0]] &&
  (IntegerQ[p] || GtQ[a,0])
```

x: $\int (a+b\,x^n)^p\,dx$ when $p\notin\mathbb{Z}^+ \bigwedge \frac{1}{n}\notin\mathbb{Z} \bigwedge \frac{1}{n}+p\notin\mathbb{Z}^- \bigwedge \neg\,(p\in\mathbb{Z}^- \bigvee a>0)$

Note: If $r == 1 \wedge (s\in\mathbb{Z} \vee t\in\mathbb{Z})$, then Hypergeometric2F1[r, s, t, z] == Hypergeometric2F1[s, r, t, z] are undefined or can be expressed in elementary form.

Note: *Mathematica* has a hard time simplifying the derivative of the following antiderivative to the integrand, so the following, more complicated, but easily differentiated, rule is used instead.

Rule 1.1.3.1.7.x: If $p\notin\mathbb{Z}^+ \bigwedge \frac{1}{n}\notin\mathbb{Z} \bigwedge \frac{1}{n}+p\notin\mathbb{Z}^- \bigwedge \neg\,(p\in\mathbb{Z}^- \bigvee a>0)$, then

$$\int (a+b\,x^n)^p\,dx \longrightarrow \frac{x\,(a+b\,x^n)^{p+1}}{a}\,\text{Hypergeometric2F1}\left[1,\,\frac{1}{n}+p+1,\,\frac{1}{n}+1,\,-\frac{b\,x^n}{a}\right]$$

Program code:

```
(* Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  x*(a+b*x^n)^(p+1)/a*Hypergeometric2F1[1,1/n+p+1,1/n+1,-b*x^n/a] /;
FreeQ[{a,b,n,p},x] && Not[IGtQ[p,0]] && Not[IntegerQ[1/n]] && Not[ILtQ[Simplify[1/n+p],0]] &&
  Not[IntegerQ[p] || GtQ[a,0]] *)
```

2: $\int (a+b\,x^n)^p\,dx$ when $p\notin\mathbb{Z}^+ \bigwedge \frac{1}{n}\notin\mathbb{Z} \bigwedge \frac{1}{n}+p\notin\mathbb{Z}^- \bigwedge \neg\,(p\in\mathbb{Z}^- \bigvee a>0)$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(a+b\,x^n)^p}{\left(1+\frac{b\,x^n}{a}\right)^p} == 0$

- **Rule 1.1.3.1.7.2: If** $p\notin\mathbb{Z}^+ \bigwedge \frac{1}{n}\notin\mathbb{Z} \bigwedge \frac{1}{n}+p\notin\mathbb{Z}^- \bigwedge \neg\,(p\in\mathbb{Z}^- \bigvee a>0)$**, then**

$$\int (a+b\,x^n)^p\,dx \longrightarrow \frac{a^{\text{IntPart}[p]}\,(a+b\,x^n)^{\text{FracPart}[p]}}{\left(1+\frac{b\,x^n}{a}\right)^{\text{FracPart}[p]}}\int\left(1+\frac{b\,x^n}{a}\right)^p dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  a^IntPart[p]*(a+b*x^n)^FracPart[p]/(1+b*x^n/a)^FracPart[p]*Int[(1+b*x^n/a)^p,x] /;
FreeQ[{a,b,n,p},x] && Not[IGtQ[p,0]] && Not[IntegerQ[1/n]] &&
  Not[ILtQ[Simplify[1/n+p],0]] && Not[IntegerQ[p] || GtQ[a,0]]
```

S: $\int (a+b\,v^n)^p\,dx$ when $v == c+d\,x$

Derivation: Integration by substitution

Rule 1.1.3.1.S: If $v == c+d\,x$**, then**

$$\int (a+b\,v^n)^p\,dx \longrightarrow \frac{1}{d}\,\text{Subst}\left[\int (a+b\,x^n)^p\,dx,\,x,\,v\right]$$

Program code:

```
Int[(a_.+b_.*v_^n_)^p_,x_Symbol] :=
  1/Coefficient[v,x,1]*Subst[Int[(a+b*x^n)^p,x],x,v] /;
FreeQ[{a,b,n,p},x] && LinearQ[v,x] && NeQ[v,x]
```

Rules for integrands of the form $(a_1 + b_1 x^n)^p (a_2 + b_2 x^n)^p$

1: $\int (a_1 + b_1 x^n)^p (a_2 + b_2 x^n)^p \, dx$ when $a_2 b_1 + a_1 b_2 == 0 \land (p \in \mathbb{Z} \lor (a_1 > 0 \land a_2 > 0))$

Derivation: Algebraic simplification

Basis: If $a_2 b_1 + a_1 b_2 == 0 \land (p \in \mathbb{Z} \lor (a_1 > 0 \land a_2 > 0))$**, then** $(a_1 + b_1 x^n)^p (a_2 + b_2 x^n)^p == \left(a_1 a_2 + b_1 b_2 x^{2n}\right)^p$

Rule: If $a_2 b_1 + a_1 b_2 == 0 \land (p \in \mathbb{Z} \lor (a_1 > 0 \land a_2 > 0))$**, then**

$$\int (a_1 + b_1 x^n)^p (a_2 + b_2 x^n)^p \, dx \longrightarrow \int \left(a_1 a_2 + b_1 b_2 x^{2n}\right)^p dx$$

Program code:

```
Int[(a1_.+b1_.*x_^n_)^p_.*(a2_.+b2_.*x_^n_)^p_.,x_Symbol] :=
  Int[(a1*a2+b1*b2*x^(2*n))^p,x] /;
FreeQ[{a1,b1,a2,b2,n,p},x] && EqQ[a2*b1+a1*b2,0] && (IntegerQ[p] || GtQ[a1,0] && GtQ[a2,0])
```

2. $\int (a_1 + b_1 x^n)^p (a_2 + b_2 x^n)^p \, dx$ when $a_2 b_1 + a_1 b_2 == 0 \land p \notin \mathbb{Z}$

1: $\int (a_1 + b_1 x^n)^p (a_2 + b_2 x^n)^p \, dx$ when $a_2 b_1 + a_1 b_2 == 0 \land p \notin \mathbb{Z} \land n \in \mathbb{Z}^+ \land p > 0$

Reference: G&R 2.110.1, CRC 88b

Derivation: Binomial recurrence 1b

Derivation: Inverted integration by parts

Note: If $n \in \mathbb{Z}^+ \land p > 0$**, then** $n p + 1 \neq 0$**.**

Rule 1.1.3.1.4.1.1.2: If $n \in \mathbb{Z}^+ \land p > 0$**, then**

$$\int (a + b x^n)^p \, dx \longrightarrow \frac{x (a + b x^n)^p}{n p + 1} + \frac{a n p}{n p + 1} \int (a + b x^n)^{p-1} \, dx$$

Program code:

```
Int[(a1_+b1_.*x_^n_.)^p_.*(a2_+b2_.*x_^n_.)^p_.,x_Symbol] :=
  x*(a1+b1*x^n)^p*(a2+b2*x^n)^p/(2*n*p+1) +
  2*a1*a2*n*p/(2*n*p+1)*Int[(a1+b1*x^n)^(p-1)*(a2+b2*x^n)^(p-1),x] /;
FreeQ[{a1,b1,a2,b2},x] && EqQ[a2*b1+a1*b2,0] && IGtQ[2*n,0] && GtQ[p,0] && (IntegerQ[2*p] || Denominator[p+1/n]<Denominator[p])
```

2: $\int (a_1 + b_1 x^n)^p (a_2 + b_2 x^n)^p \, dx$ when $a_2 b_1 + a_1 b_2 == 0 \land p \notin \mathbb{Z} \land n \in \mathbb{Z}^+ \land p < -1$

Reference: G&R 2.110.2, CRC 88d

Derivation: Binomial recurrence 2b

Derivation: Integration by parts

- **Basis:** $(a + b x^n)^p == x^{n (p+1)+1} \frac{(a+b x^n)^p}{x^{n (p+1)+1}}$
- **Basis:** $\int \frac{(a+b x^n)^p}{x^{n (p+1)+1}} dx == -\frac{(a+b x^n)^{p+1}}{x^{n (p+1)} a n (p+1)}$

Rule 1.1.3.1.4.1.2: If $n \in \mathbb{Z}^+ \land p < -1$**, then**

$$\int (a + b x^n)^p \, dx \longrightarrow -\frac{x (a + b x^n)^{p+1}}{a n (p + 1)} + \frac{n (p + 1) + 1}{a n (p + 1)} \int (a + b x^n)^{p+1} \, dx$$

Program code:

```
Int[(a1_+b1_.*x_^n_.)^p_*(a2_+b2_.*x_^n_.)^p_,x_Symbol] :=
  -x*(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1)/(2*a1*a2*n*(p+1)) +
  (2*n*(p+1)+1)/(2*a1*a2*n*(p+1))*Int[(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1),x] /;
FreeQ[{a1,b1,a2,b2},x] && EqQ[a2*b1+a1*b2,0] && IGtQ[2*n,0] && LtQ[p,-1] && (IntegerQ[2*p] || Denominator[p+1/n]<Denominator[p])
```

3: $\int (a_1 + b_1 x^n)^p (a_2 + b_2 x^n)^p \, dx$ when $a_2 b_1 + a_1 b_2 == 0 \land p \notin \mathbb{Z} \land n \in \mathbb{Z}^-$

Derivation: Integration by substitution

- **Basis: If** $n \in \mathbb{Z}$**, then** $F[x^n] == -\text{Subst}\left[\frac{F[x^{-n}]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Rule 1.1.3.1.4.2: If $n \in \mathbb{Z}^-$**, then**

$$\int (a + b x^n)^p \, dx \longrightarrow -\text{Subst}\left[\int \frac{(a + b x^{-n})^p}{x^2} dx, x, \frac{1}{x}\right]$$

Program code:

```
Int[(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  -Subst[Int[(a1+b1*x^(-n))^p*(a2+b2*x^(-n))^p/x^2,x],x,1/x] /;
FreeQ[{a1,b1,a2,b2,p},x] && EqQ[a2*b1+a1*b2,0] && ILtQ[2*n,0]
```

4: $\int (a_1 + b_1 x^n)^p (a_2 + b_2 x^n)^p \, dx$ when $a_2 b_1 + a_1 b_2 == 0 \wedge p \notin \mathbb{Z} \wedge n \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, **then** $F[x^n] == k \, \text{Subst}\left[x^{k-1} F\left[x^{k\,n}\right], x, x^{1/k}\right] \partial_x x^{1/k}$

Rule 1.1.3.1.5: If $n \notin \mathbb{Z} \wedge n \in \mathbb{F}$, **let** $k = \text{Denominator}[n]$, **then**

$$\int (a + b x^n)^p \, dx \longrightarrow k \, \text{Subst}\left[\int x^{k-1} \left(a + b x^{k\,n}\right)^p dx, x, x^{1/k}\right]$$

Program code:

```
Int[(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  With[{k=Denominator[2*n]},
  k*Subst[Int[x^(k-1)*(a1+b1*x^(k*n))^p*(a2+b2*x^(k*n))^p,x],x,x^(1/k)]] /;
FreeQ[{a1,b1,a2,b2,p},x] && EqQ[a2*b1+a1*b2,0] && FractionQ[2*n]
```

5: $\int (a_1 + b_1 x^n)^p (a_2 + b_2 x^n)^p \, dx$ when $a_2 b_1 + a_1 b_2 == 0 \wedge p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $a_2 b_1 + a_1 b_2 == 0$, **then** $\partial_x \frac{(a_1+b_1 x^n)^p (a_2+b_2 x^n)^p}{(a_1 a_2+b_1 b_2 x^{2n})^p} == 0$

Rule: If $a_2 b_1 + a_1 b_2 == 0 \wedge p \notin \mathbb{Z}$, **then**

$$\int (a_1 + b_1 x^n)^p (a_2 + b_2 x^n)^p \, dx \longrightarrow \frac{(a_1 + b_1 x^n)^{\text{FracPart}[p]} (a_2 + b_2 x)^{\text{FracPart}[p]}}{\left(a_1 a_2 + b_1 b_2 x^2\right)^{\text{FracPart}[p]}} \int \left(a_1 a_2 + b_1 b_2 x^{2n}\right)^p dx$$

Program code:

```
Int[(a1_.+b1_.*x_^n_)^p_*(a2_.+b2_.*x_^n_)^p_,x_Symbol] :=
  (a1+b1*x^n)^FracPart[p]*(a2+b2*x^n)^FracPart[p]/(a1*a2+b1*b2*x^(2*n))^FracPart[p]*Int[(a1*a2+b1*b2*x^(2*n))^p,x] /;
FreeQ[{a1,b1,a2,b2,n,p},x] && EqQ[a2*b1+a1*b2,0] && Not[IntegerQ[p]]
```

Rules for integrands of the form $(a + b (c x^q)^n)^p$

1: $\int (a + b (c x^q)^n)^p \, dx$ when $n q \in \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x \frac{(d x)^{m+1}}{\left((c x^q)^{1/q}\right)^{m+1}} == 0$
- **Basis:** $\frac{F\left[(c x^q)^{1/q}\right]}{x} == \text{Subst}\left[\frac{F[x]}{x}, x, (c x^q)^{1/q}\right] \partial_x (c x^q)^{1/q}$

Rule: If $n q \in \mathbb{Z}$**, then**

$$\int (a + b (c x^q)^n)^p \, dx \longrightarrow \frac{x}{(c x^q)^{1/q}} \int \frac{(c x^q)^{1/q} \left(a + b \left((c x^q)^{1/q}\right)^{n q}\right)^p}{x} \, dx$$

$$\longrightarrow \frac{x}{(c x^q)^{1/q}} \text{Subst}\left[\int (a + b x^{n q})^p \, dx, x, (c x^q)^{1/q}\right]$$

Program code:

```
Int[(a_+b_.*(c_.*x_^q_.)^n_)^p_.,x_Symbol] :=
  x/(c*x^q)^(1/q)*Subst[Int[(a+b*x^(n*q))^p,x],x,(c*x^q)^(1/q)] /;
FreeQ[{a,b,c,n,p,q},x] && IntegerQ[n*q] && NeQ[x,(c*x^q)^(1/q)]
```

2: $\int (a + b (c x^q)^n)^p \, dx$ when $n \in \mathbb{F}$

Derivation: Integration by substitution

Rule 1.1.3.2.S.4.3: If $n \in \mathbb{F}$**, then**

$$\int (a + b (c x^q)^n)^p \, dx \longrightarrow \text{Subst}\left[\int (a + b c^n x^{n q})^p \, dx, x^{1/k}, \frac{(c x^q)^{1/k}}{c^{1/k} \left(x^{1/k}\right)^{q-1}}\right]$$

Program code:

```
Int[(a_+b_.*(c_.*x_^q_.)^n_)^p_.,x_Symbol] :=
  With[{k=Denominator[n]},
  Subst[Int[(a+b*c^n*x^(n*q))^p,x],x^(1/k),(c*x^q)^(1/k)/(c^(1/k)*(x^(1/k))^(q-1))]] /;
FreeQ[{a,b,c,p,q},x] && FractionQ[n]
```

3: $\int (a+b\,(c\,x^q)^n)^p\,dx$ when $n \notin \mathbb{R}$

Derivation: Integration by substitution

- **Basis:** $F[(c\,x^q)^n] = \text{Subst}\left[F[c^n\,x^{n\,q}],\,x^{n\,q},\,\frac{(c\,x^q)^n}{c^n}\right]$
- **Rule: If** $n \notin \mathbb{R}$**, then**

$$\int (a+b\,(c\,x^q)^n)^p\,dx \;\longrightarrow\; \text{Subst}\left[\int (a+b\,c^n\,x^{n\,q})^p\,dx,\,x^{n\,q},\,\frac{(c\,x^q)^n}{c^n}\right]$$

- **Program code:**

```
Int[(a_+b_.*(c_.*x_^q_.)^n_)^p_.,x_Symbol] :=
  Subst[Int[(a+b*c^n*x^(n*q))^p,x],x^(n*q),(c*x^q)^n/c^n] /;
FreeQ[{a,b,c,n,p,q},x] && Not[RationalQ[n]]
```

?: $\int (a+b\,v^n)^p\,dx$ when $v == d\,x^q \wedge q \in \mathbb{Z}^-$

Derivation: Integration by substitution

- **Basis: If** $q \in \mathbb{Z}$**, then** $F[x^q] = -\text{Subst}\left[\frac{F[x^{-q}]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x \frac{1}{x}$

Rule 1.1.3.1.S.2.2: If $v == d\,x^q \wedge q \in \mathbb{Z}^-$**, then**

$$\int (a+b\,v^n)^p\,dx \;\longrightarrow\; -\text{Subst}\left[\int \frac{(a+b\,(d\,x^{-q})^n)^p}{x^2}\,dx,\,x,\,\frac{1}{x}\right]$$

- **Program code:**

```
Int[(a_+b_.*(d_.*x_^q_.)^n_)^p_.,x_Symbol] :=
  -Subst[Int[(a+b*(d*x^(-q))^n)^p/x^2,x],x,1/x] /;
FreeQ[{a,b,d,n,p},x] && ILtQ[q,0]
```

2: $\int (a+b\,v^n)^p\,dx$ when $v == d\,x^q \wedge q \in \mathbb{F}$

- **Derivation: Integration by substitution**
- **Basis: If** $s \in \mathbb{Z}^+$**, then** $F\left[x^{1/s}\right] == s\,\text{Subst}\left[x^{s-1}\,F[x],\,x,\,x^{1/s}\right]\,\partial_x x^{1/s}$
- **Rule 1.1.3.1.S.2.2: If** $v == d\,x^q \wedge q \in \mathbb{F}$**, let** $s \to \text{Denominator}[q]$**, then**

$$\int (a+b\,v^n)^p\,dx \longrightarrow \int (a+b\,v^n)^p\,dx \longrightarrow \int \left(a+b\left(d\left(x^{1/s}\right)^{q\,s}\right)^n\right)^p dx \longrightarrow s\,\text{Subst}\left[\int x^{s-1}\,(a+b\,(d\,x^{q\,s})^n)^p\,dx,\,x,\,x^{1/s}\right]$$

- **Program code:**

```
Int[(a_+b_.*(d_.*x_^q_.)^n_)^p_.,x_Symbol] :=
  With[{s=Denominator[q]},
  s*Subst[Int[x^(s-1)*(a+b*(d*x^(q*s))^n)^p,x],x,x^(1/s)]] /;
FreeQ[{a,b,d,n,p},x] && FractionQ[q]
```

x: $\int (a+b\,v^n)^p\,dx$ when $v == d\,x^q \wedge n\,q \notin \mathbb{Z}$

- **Derivation: Integration by substitution**
- **Rule 1.1.3.1.S.2.3: If** $v == d\,x^q \wedge n\,q \notin \mathbb{Z}$**, then**

$$\int (a+b\,v^n)^p\,dx \longrightarrow \text{Subst}\left[\int (a+b\,x^{n\,q})^p\,dx,\,x^{n\,q},\,v^n\right]$$

- **Program code:**

```
(* Int[(a_+b_.*(d_.*x_^q_.)^n_)^p_.,x_Symbol] :=
  Subst[Int[(a+b*x^(n*q))^p,x],x^(n*q),(d*x^q)^n] /;
FreeQ[{a,b,d,n,p,q},x] && Not[IntegerQ[n*q]] && NeQ[x^(n*q),(d*x^q)^n] *)
```