Rules for integrands of the form $(a + b \sin[e + f x])^m (c + d \sin[e + f x])^n$

1: $\int (a + b \sin[e + f x]) (c + d \sin[e + f x]) \, dx$ when $b c - a d \neq 0$

- **Derivation: Algebraic expansion**
- **Basis:** $(a + b z) (c + d z) == \frac{1}{2} (2 a c + b d) + (b c + a d) z - \frac{1}{2} b d \left(1 - 2 z^2\right)$
- **Rule: If** $b c - a d \neq 0$**, then**

$$\int (a + b \sin[e + f x]) (c + d \sin[e + f x]) \, dx \longrightarrow \frac{(2 a c + b d) x}{2} - \frac{(b c + a d) \cos[e + f x]}{f} - \frac{b d \cos[e + f x] \sin[e + f x]}{2 f}$$

- **Program code:**

```
Int[(a_+b_.*sin[e_.+f_.*x_])*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  (2*a*c+b*d)*x/2 - (b*c+a*d)*Cos[e+f*x]/f - b*d*Cos[e+f*x]*Sin[e+f*x]/(2*f) /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0]
```

2: $\int \frac{a + b \sin[e + f x]}{c + d \sin[e + f x]} \, dx$ when $b c - a d \neq 0$

- **Reference: G&R 2.551.2**
- **Derivation: Algebraic expansion**
- **Basis:** $\frac{a+b z}{c+d z} == \frac{b}{d} - \frac{b c-a d}{d (c+d z)}$
- **Rule: If** $b c - a d \neq 0$**, then**

$$\int \frac{a + b \sin[e + f x]}{c + d \sin[e + f x]} \, dx \longrightarrow \frac{b x}{d} - \frac{b c - a d}{d} \int \frac{1}{c + d \sin[e + f x]} \, dx$$

- **Program code:**

```
Int[(a_.+b_.*sin[e_.+f_.*x_])/(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  b*x/d - (b*c-a*d)/d*Int[1/(c+d*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0]
```

3. $\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n \, dx$ when $b c + a d == 0 \land a^2 - b^2 == 0$

1: $\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n \, dx$ when $b c + a d == 0 \land a^2 - b^2 == 0 \land m \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b c + a d == 0 \land a^2 - b^2 == 0$**, then** $(a + b \operatorname{Sin}[z]) (c + d \operatorname{Sin}[z]) == a c \operatorname{Cos}[z]^2$

Rule: If $b c + a d == 0 \land a^2 - b^2 == 0 \land m \in \mathbb{Z}$**, then**

$$\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n \, dx \longrightarrow a^m c^m \int \operatorname{Cos}[e + f x]^{2 m} (c + d \operatorname{Sin}[e + f x])^{n-m} \, dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  a^m*c^m*Int[Cos[e+f*x]^(2*m)*(c+d*Sin[e+f*x])^(n-m),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[m] && Not[IntegerQ[n] && (LtQ[m,0] && GtQ[n,0] || LtQ[0,n,
```

2. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\ \wedge\ a^2-b^2==0\ \wedge\ m\notin\mathbb{Z}\ \wedge\ n\notin\mathbb{Z}$

   1. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\ \bigwedge\ a^2-b^2==0\ \bigwedge\ m+\frac{1}{2}\in\mathbb{Z}^+$

      1. $\int \sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\ \wedge\ a^2-b^2==0$

         1: $\int \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c+a\,d==0\ \wedge\ a^2-b^2==0$

Derivation: Piecewise constant extraction

Basis: If $b\,c+a\,d==0\ \wedge\ a^2-b^2==0$**, then** $\partial_x \frac{\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\ \sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}==0$

Rule: If $b\,c+a\,d==0\ \wedge\ a^2-b^2==0$**, then**

$$\int \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx \longrightarrow \frac{a\,c\,\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\ \sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\int \frac{\mathrm{Cos}[e+f\,x]}{c+d\,\mathrm{Sin}[e+f\,x]}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/Sqrt[c_+d_.*sin[e_.+f_.*x_]],x_Symbol] :=
  a*c*Cos[e+f*x]/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]])*Int[Cos[e+f*x]/(c+d*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0]
```

         2: $\int \sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\ \bigwedge\ a^2-b^2==0\ \bigwedge\ n\neq-\frac{1}{2}$

Derivation: Doubly degenerate sine recurrence 1a with $p\to 0$

Rule: If $b\,c+a\,d==0\ \bigwedge\ a^2-b^2==0\ \bigwedge\ n\neq-\frac{1}{2}$**, then**

$$\int \sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx \longrightarrow -\frac{2\,b\,\mathrm{Cos}[e+f\,x]\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{f\,(2\,n+1)\,\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -2*b*Cos[e+f*x]*(c+d*Sin[e+f*x])^n/(f*(2*n+1)*Sqrt[a+b*Sin[e+f*x]]) /;
FreeQ[{a,b,c,d,e,f,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && NeQ[n,-1/2]
```

2. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d=0\,\wedge\,a^2-b^2=0\,\wedge\,m-\frac{1}{2}\in\mathbb{Z}^+$

1: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d=0\,\wedge\,a^2-b^2=0\,\wedge\,m-\frac{1}{2}\in\mathbb{Z}^+\,\wedge\,n<-1$

Derivation: Doubly degenerate sine recurrence 1a with $p\to 0$

Rule: If $b\,c+a\,d=0\,\wedge\,a^2-b^2=0\,\wedge\,m-\frac{1}{2}\in\mathbb{Z}^+\,\wedge\,n<-1$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx\,\longrightarrow$$

$$-\frac{2\,b\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^n}{f\,(2\,n+1)}-\frac{b\,(2\,m-1)}{d\,(2\,n+1)}\int (a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^{n+1}\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -2*b*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^n/(f*(2*n+1)) -
  b*(2*m-1)/(d*(2*n+1))*Int[(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IGtQ[m-1/2,0] && LtQ[n,-1] && Not[ILtQ[m+n,0] && GtQ[2*m+n+1,0]]
```

2: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d=0\,\wedge\,a^2-b^2=0\,\wedge\,m-\frac{1}{2}\in\mathbb{Z}^+\,\wedge\,n\nless -1$

Derivation: Doubly degenerate sine recurrence 1b with $p\to 0$

Rule: If $b\,c+a\,d=0\,\wedge\,a^2-b^2=0\,\wedge\,m-\frac{1}{2}\in\mathbb{Z}^+\,\wedge\,n\nless -1$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx\,\longrightarrow$$

$$-\frac{b\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^n}{f\,(m+n)}+\frac{a\,(2\,m-1)}{m+n}\int (a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -b*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^n/(f*(m+n)) +
  a*(2*m-1)/(m+n)*Int[(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IGtQ[m-1/2,0] && Not[LtQ[n,-1]] &&
  Not[IGtQ[n-1/2,0] && LtQ[n,m]] && Not[ILtQ[m+n,0] && GtQ[2*m+n+1,0]]
```

2. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\ \wedge\ a^2-b^2==0\ \wedge\ m+n\in\mathbb{Z}^-$

1. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\ \wedge\ a^2-b^2==0\ \wedge\ m+n+1==0$

1: $\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c+a\,d==0\ \wedge\ a^2-b^2==0$

Derivation: Piecewise constant extraction

Basis: If $b\,c+a\,d==0\ \wedge\ a^2-b^2==0$**, then** $\partial_x \frac{\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}==0$

Rule: If $b\,c+a\,d==0\ \wedge\ a^2-b^2==0$**, then**

$$\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx \longrightarrow \frac{\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\int \frac{1}{\mathrm{Cos}[e+f\,x]}\,dx$$

Program code:

```
Int[1/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  Cos[e+f*x]/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]])*Int[1/Cos[e+f*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0]
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\ \bigwedge\ a^2-b^2==0\ \bigwedge\ m+n+1==0\ \bigwedge\ m\neq-\frac{1}{2}$

Derivation: Doubly degenerate sine recurrence 1c with $n\to-m-1,\ p\to0$

Rule: If $b\,c+a\,d==0\ \bigwedge\ a^2-b^2==0\ \bigwedge\ m+n+1==0\ \bigwedge\ m\neq-\frac{1}{2}$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx \longrightarrow \frac{b\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{a\,f\,(2\,m+1)}$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  b*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n/(a*f*(2*m+1)) /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && EqQ[m+n+1,0] && NeQ[m,-1/2]
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\bigwedge\,a^2-b^2==0\,\bigwedge\,m+n+1\in\mathbb{Z}^-\,\bigwedge\,m\neq-\frac{1}{2}$

Derivation: Doubly degenerate sine recurrence 1c with $p\to 0$

Rule: If $b\,c+a\,d==0\,\bigwedge\,a^2-b^2==0\,\bigwedge\,m+n+1\in\mathbb{Z}^-\,\bigwedge\,m\neq-\frac{1}{2}$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\longrightarrow$$
$$\frac{b\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{a\,f\,(2\,m+1)}+\frac{m+n+1}{a\,(2\,m+1)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  b*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n/(a*f*(2*m+1)) +
  (m+n+1)/(a*(2*m+1))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && ILtQ[Simplify[m+n+1],0] && NeQ[m,-1/2] &&
  (SumSimplerQ[m,1] || Not[SumSimplerQ[n,1]])
```

3: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\wedge\,a^2-b^2==0\,\wedge\,m<-1$

Derivation: Doubly degenerate sine recurrence 1c with $p\to 0$

Rule: If $b\,c+a\,d==0\,\wedge\,a^2-b^2==0\,\wedge\,m<-1$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\longrightarrow$$
$$\frac{b\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{a\,f\,(2\,m+1)}+\frac{m+n+1}{a\,(2\,m+1)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  b*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n/(a*f*(2*m+1)) +
  (m+n+1)/(a*(2*m+1))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && LtQ[m,-1] && Not[LtQ[m,n,-1]] && IntegersQ[2*m,2*n]
```

4: $\int (a + b\,\text{Sin}[e + f\,x])^m\,(c + d\,\text{Sin}[e + f\,x])^n\,dx$ when $b\,c + a\,d == 0\,\wedge\,a^2 - b^2 == 0\,\wedge\,m \notin \mathbb{Z}\,\wedge\,n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis: If** $b\,c + a\,d == 0\,\wedge\,a^2 - b^2 == 0$, **then** $\partial_x \frac{(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^m}{\text{Cos}[e+f\,x]^{2\,m}} == 0$

- **Rule: If** $b\,c + a\,d == 0\,\wedge\,a^2 - b^2 == 0\,\wedge\,m \notin \mathbb{Z}\,\wedge\,n \notin \mathbb{Z}$, **then**

$$\int (a + b\,\text{Sin}[e + f\,x])^m\,(c + d\,\text{Sin}[e + f\,x])^n\,dx \longrightarrow$$

$$\left(a^{\text{IntPart}[m]}\,c^{\text{IntPart}[m]}\,(a + b\,\text{Sin}[e + f\,x])^{\text{FracPart}[m]}\,(c + d\,\text{Sin}[e + f\,x])^{\text{FracPart}[m]}\right) \Big/ \text{Cos}[e + f\,x]^{2\,\text{FracPart}[m]} \int \text{Cos}[e + f\,x]^{2\,m}\,(c + d\,\text{Sin}[e + f\,x])^{n-m}\,dx$$

- **Program code:**

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  a^IntPart[m]*c^IntPart[m]*(a+b*Sin[e+f*x])^FracPart[m]*(c+d*Sin[e+f*x])^FracPart[m]/Cos[e+f*x]^(2*FracPart[m])*
    Int[Cos[e+f*x]^(2*m)*(c+d*Sin[e+f*x])^(n-m),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && (FractionQ[m] || Not[FractionQ[n]])
```

4: $\int \frac{(a + b\,\text{Sin}[e + f\,x])^2}{c + d\,\text{Sin}[e + f\,x]}\,dx$ when $b\,c - a\,d \neq 0$

Derivation: Algebraic expansion

- **Basis:** $\frac{(a+b\,z)^2}{c+d\,z} == \frac{b^2\,z}{d} + \frac{a^2\,d-b\,(b\,c-2\,a\,d)\,z}{d\,(c+d\,z)}$

- **Rule: If** $b\,c - a\,d \neq 0$, **then**

$$\int \frac{(a + b\,\text{Sin}[e + f\,x])^2}{c + d\,\text{Sin}[e + f\,x]}\,dx \longrightarrow -\frac{b^2\,\text{Cos}[e + f\,x]}{d\,f} + \frac{1}{d} \int \frac{a^2\,d - b\,(b\,c - 2\,a\,d)\,\text{Sin}[e + f\,x]}{c + d\,\text{Sin}[e + f\,x]}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^2/(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -b^2*Cos[e+f*x]/(d*f) + 1/d*Int[Simp[a^2*d-b*(b*c-2*a*d)*Sin[e+f*x],x]/(c+d*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0]
```

5: $\int \frac{1}{(a+b\,\text{Sin}[e+f\,x])\,(c+d\,\text{Sin}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{(a+b\,z)\,(c+d\,z)} == \frac{b}{(b\,c-a\,d)\,(a+b\,z)} - \frac{d}{(b\,c-a\,d)\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0$**, then**

$$\int \frac{1}{(a+b\,\text{Sin}[e+f\,x])\,(c+d\,\text{Sin}[e+f\,x])}\,dx \longrightarrow \frac{b}{b\,c-a\,d}\int \frac{1}{a+b\,\text{Sin}[e+f\,x]}\,dx - \frac{d}{b\,c-a\,d}\int \frac{1}{c+d\,\text{Sin}[e+f\,x]}\,dx$$

Program code:

```
Int[1/((a_.+b_.*sin[e_.+f_.*x_])*(c_.+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  b/(b*c-a*d)*Int[1/(a+b*Sin[e+f*x]),x] - d/(b*c-a*d)*Int[1/(c+d*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0]
```

6. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0$

1: $\int (b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,dx$

Derivation: Algebraic expansion

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0$**, then**

$$\int (b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,dx \longrightarrow c\int (b\,\text{Sin}[e+f\,x])^m\,dx + \frac{d}{b}\int (b\,\text{Sin}[e+f\,x])^{m+1}\,dx$$

Program code:

```
Int[(b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  c*Int[(b*Sin[e+f*x])^m,x] + d/b*Int[(b*Sin[e+f*x])^(m+1),x] /;
FreeQ[{b,c,d,e,f,m},x]
```

2. $\int (a + b \sin[e + f x])^m (c + d \sin[e + f x]) \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 == 0$

1: $\int (a + b \sin[e + f x])^m (c + d \sin[e + f x]) \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 == 0 \land a d m + b c (m + 1) == 0$

- **Derivation: Singly degenerate sine recurrence 2a with** $A \to -\frac{a d m}{b (m+1)}, \; B \to d, \; n \to 0, \; p \to 0$
- **Derivation: Singly degenerate sine recurrence 2c with** $A \to -\frac{a d m}{b (m+1)}, \; B \to d, \; n \to 0, \; p \to 0$
- **Note: If** $a^2 - b^2 == 0 \land a d m + b c (m + 1) == 0$**, then** $m + 1 \neq 0$**.**
- **Rule: If** $b c - a d \neq 0 \land a^2 - b^2 == 0 \land a d m + b c (m + 1) == 0$**, then**

$$\int (a + b \sin[e + f x])^m (c + d \sin[e + f x]) \, dx \longrightarrow -\frac{d \cos[e + f x] (a + b \sin[e + f x])^m}{f (m + 1)}$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -d*Cos[e+f*x]*(a+b*Sin[e+f*x])^m/(f*(m+1)) /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && EqQ[a*d*m+b*c*(m+1),0]
```

2: $\int (a + b \sin[e + f x])^m (c + d \sin[e + f x]) \, dx$ when $b c - a d \neq 0 \bigwedge a^2 - b^2 == 0 \bigwedge m < -\frac{1}{2}$

- **Derivation: Singly degenerate sine recurrence 2a with** $A \to c, \; B \to d, \; n \to 0, \; p \to 0$
- **Rule: If** $b c - a d \neq 0 \bigwedge a^2 - b^2 == 0 \bigwedge m < -\frac{1}{2}$**, then**

$$\int (a + b \sin[e + f x])^m (c + d \sin[e + f x]) \, dx \longrightarrow$$
$$\frac{(b c - a d) \cos[e + f x] (a + b \sin[e + f x])^m}{a f (2 m + 1)} + \frac{a d m + b c (m + 1)}{a b (2 m + 1)} \int (a + b \sin[e + f x])^{m+1} \, dx$$

- **Program code:**

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  (b*c-a*d)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m/(a*f*(2*m+1)) +
  (a*d*m+b*c*(m+1))/(a*b*(2*m+1))*Int[(a+b*Sin[e+f*x])^(m+1),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && LtQ[m,-1/2]
```

3: $\int (a + b \sin[e + f x])^m (c + d \sin[e + f x]) \, dx$ when $b c - a d \neq 0 \bigwedge a^2 - b^2 == 0 \bigwedge m \nless -\frac{1}{2}$

Derivation: Singly degenerate sine recurrence 2c with $A \to c,\ B \to d,\ n \to 0,\ p \to 0$

Rule: If $b c - a d \neq 0 \bigwedge a^2 - b^2 == 0 \bigwedge m \nless -\frac{1}{2}$**, then**

$$\int (a + b \sin[e + f x])^m (c + d \sin[e + f x]) \, dx \longrightarrow$$
$$-\frac{d \cos[e + f x] (a + b \sin[e + f x])^m}{f (m + 1)} + \frac{a d m + b c (m + 1)}{b (m + 1)} \int (a + b \sin[e + f x])^m \, dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -d*Cos[e+f*x]*(a+b*Sin[e+f*x])^m/(f*(m+1)) +
  (a*d*m+b*c*(m+1))/(b*(m+1))*Int[(a+b*Sin[e+f*x])^m,x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1/2]]
```

3. $\int (a + b \sin[e + f x])^m (c + d \sin[e + f x]) \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0$

1. $\int (a + b \sin[e + f x])^m (c + d \sin[e + f x]) \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land 2 m \in \mathbb{Z}$

1: $\int \frac{c + d \sin[e + f x]}{\sqrt{a + b \sin[e + f x]}} \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $c + d z == \frac{b c - a d}{b} + \frac{d}{b} (a + b z)$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 \neq 0$**, then**

$$\int \frac{c + d \sin[e + f x]}{\sqrt{a + b \sin[e + f x]}} \, dx \longrightarrow \frac{b c - a d}{b} \int \frac{1}{\sqrt{a + b \sin[e + f x]}} \, dx + \frac{d}{b} \int \sqrt{a + b \sin[e + f x]} \, dx$$

Program code:

```
Int[(c_.+d_.*sin[e_.+f_.*x_])/Sqrt[a_+b_.*sin[e_.+f_.*x_]],x_Symbol] :=
  (b*c-a*d)/b*Int[1/Sqrt[a+b*Sin[e+f*x]],x] + d/b*Int[Sqrt[a+b*Sin[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0]
```

2: $\int (a + b\,\text{Sin}[e + f\,x])^m\,(c + d\,\text{Sin}[e + f\,x])\,dx$ when $b\,c - a\,d \neq 0\,\wedge\,a^2 - b^2 \neq 0\,\wedge\,m > 0\,\wedge\,2\,m \in \mathbb{Z}$

Reference: G&R 2.551.1 inverted

Derivation: Nondegenerate sine recurrence 1b with $A \to a\,c,\ B \to b\,c + a\,d,\ C \to b\,d,\ m \to 0,\ n \to n - 1,\ p \to 0$

Rule: If $b\,c - a\,d \neq 0\,\wedge\,a^2 - b^2 \neq 0\,\wedge\,m > 0\,\wedge\,2\,m \in \mathbb{Z}$**, then**

$$\int (a + b\,\text{Sin}[e + f\,x])^m\,(c + d\,\text{Sin}[e + f\,x])\,dx \longrightarrow$$
$$-\frac{d\,\text{Cos}[e + f\,x]\,(a + b\,\text{Sin}[e + f\,x])^m}{f\,(m + 1)} + \frac{1}{m + 1}\int (a + b\,\text{Sin}[e + f\,x])^{m-1}\,(b\,d\,m + a\,c\,(m + 1) + (a\,d\,m + b\,c\,(m + 1))\,\text{Sin}[e + f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -d*Cos[e+f*x]*(a+b*Sin[e+f*x])^m/(f*(m+1)) +
  1/(m+1)*Int[(a+b*Sin[e+f*x])^(m-1)*Simp[b*d*m+a*c*(m+1)+(a*d*m+b*c*(m+1))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && GtQ[m,0] && IntegerQ[2*m]
```

3: $\int (a + b\,\text{Sin}[e + f\,x])^m\,(c + d\,\text{Sin}[e + f\,x])\,dx$ when $b\,c - a\,d \neq 0\,\wedge\,a^2 - b^2 \neq 0\,\wedge\,m < -1\,\wedge\,2\,m \in \mathbb{Z}$

Reference: G&R 2.551.1

Derivation: Nondegenerate sine recurrence 1a with $A \to c,\ B \to d,\ C \to 0,\ n \to 0,\ p \to 0$

Rule: If $b\,c - a\,d \neq 0\,\wedge\,a^2 - b^2 \neq 0\,\wedge\,m < -1\,\wedge\,2\,m \in \mathbb{Z}$**, then**

$$\int (a + b\,\text{Sin}[e + f\,x])^m\,(c + d\,\text{Sin}[e + f\,x])\,dx \longrightarrow$$
$$-\frac{(b\,c - a\,d)\,\text{Cos}[e + f\,x]\,(a + b\,\text{Sin}[e + f\,x])^{m+1}}{f\,(m + 1)\,\left(a^2 - b^2\right)} + \frac{1}{(m + 1)\,\left(a^2 - b^2\right)}\int (a + b\,\text{Sin}[e + f\,x])^{m+1}\,((a\,c - b\,d)\,(m + 1) - (b\,c - a\,d)\,(m + 2)\,\text{Sin}[e + f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -(b*c-a*d)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(f*(m+1)*(a^2-b^2)) +
  1/((m+1)*(a^2-b^2))*Int[(a+b*Sin[e+f*x])^(m+1)*Simp[(a*c-b*d)*(m+1)-(b*c-a*d)*(m+2)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && LtQ[m,-1] && IntegerQ[2*m]
```

2. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,2\,m\notin\mathbb{Z}$

1: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,2\,m\notin\mathbb{Z}\,\wedge\,c^2-d^2==0$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x \frac{\mathrm{Cos}[e+f\,x]}{\sqrt{1+\mathrm{Sin}[e+f\,x]}\,\sqrt{1-\mathrm{Sin}[e+f\,x]}}==0$

- **Basis:** $\mathrm{Cos}[e+f\,x]==\frac{1}{f}\,\partial_x\mathrm{Sin}[e+f\,x]$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,2\,m\notin\mathbb{Z}\,\wedge\,c^2-d^2==0$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{c\,\mathrm{Cos}[e+f\,x]}{\sqrt{1+\mathrm{Sin}[e+f\,x]}\,\sqrt{1-\mathrm{Sin}[e+f\,x]}}\int\frac{\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,\sqrt{1+\frac{d}{c}\,\mathrm{Sin}[e+f\,x]}}{\sqrt{1-\frac{d}{c}\,\mathrm{Sin}[e+f\,x]}}\,dx\,\longrightarrow$$

$$\frac{c\,\mathrm{Cos}[e+f\,x]}{f\,\sqrt{1+\mathrm{Sin}[e+f\,x]}\,\sqrt{1-\mathrm{Sin}[e+f\,x]}}\,\mathrm{Subst}\Big[\int\frac{(a+b\,x)^m\,\sqrt{1+\frac{d}{c}\,x}}{\sqrt{1-\frac{d}{c}\,x}}\,dx,\,x,\,\mathrm{Sin}[e+f\,x]\Big]$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  c*Cos[e+f*x]/(f*Sqrt[1+Sin[e+f*x]]*Sqrt[1-Sin[e+f*x]])*Subst[Int[(a+b*x)^m*Sqrt[1+d/c*x]/Sqrt[1-d/c*x],x],x,Sin[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && Not[IntegerQ[2*m]] && EqQ[c^2-d^2,0]
```

2: $\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x]) \, dx$ when $b c - a d \neq 0 \wedge a^2 - b^2 \neq 0 \wedge 2 m \notin \mathbb{Z} \wedge c^2 - d^2 \neq 0$

Derivation: Algebraic expansion

Basis: $c + d z == \frac{b c - a d}{b} + \frac{d}{b} (a + b z)$

Rule: If $b c - a d \neq 0 \wedge a^2 - b^2 \neq 0$**, then**

$$\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x]) \, dx \longrightarrow \frac{b c - a d}{b} \int (a + b \operatorname{Sin}[e + f x])^m \, dx + \frac{d}{b} \int (a + b \operatorname{Sin}[e + f x])^{m+1} \, dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  (b*c-a*d)/b*Int[(a+b*Sin[e+f*x])^m,x] + d/b*Int[(a+b*Sin[e+f*x])^(m+1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0]
```

7. $\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n \, dx$ when $b c - a d \neq 0 \wedge a^2 - b^2 == 0 \wedge c^2 - d^2 \neq 0$

1: $\int (a + b \operatorname{Sin}[e + f x])^m (d \operatorname{Sin}[e + f x])^n \, dx$ when $a^2 - b^2 == 0 \wedge m \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $a^2 - b^2 == 0 \wedge m \in \mathbb{Z}^+$**, then**

$$\int (a + b \operatorname{Sin}[e + f x])^m (d \operatorname{Sin}[e + f x])^n \, dx \longrightarrow \int \operatorname{ExpandTrig}[(a + b \operatorname{Sin}[e + f x])^m (d \operatorname{Sin}[e + f x])^n, x] \, dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[ExpandTrig[(a+b*sin[e+f*x])^m*(d*sin[e+f*x])^n,x],x] /;
FreeQ[{a,b,d,e,f,n},x] && EqQ[a^2-b^2,0] && IGtQ[m,0] && RationalQ[n]
```

2. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^2\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0$

1. $\int \mathrm{Sin}[e+f\,x]^2\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0$

1: $\int \mathrm{Sin}[e+f\,x]^2\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\,\bigwedge\,m<-\frac{1}{2}$

Derivation: ???

Rule: If $a^2-b^2==0\,\bigwedge\,m<-\frac{1}{2}$, **then**

$$\int \mathrm{Sin}[e+f\,x]^2\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx\,\longrightarrow$$
$$\frac{b\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^m}{a\,f\,(2\,m+1)}-\frac{1}{a^2\,(2\,m+1)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(a\,m-b\,(2\,m+1)\,\mathrm{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[sin[e_.+f_.*x_]^2*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  b*Cos[e+f*x]*(a+b*Sin[e+f*x])^m/(a*f*(2*m+1)) -
  1/(a^2*(2*m+1))*Int[(a+b*Sin[e+f*x])^(m+1)*(a*m-b*(2*m+1)*Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f},x] && EqQ[a^2-b^2,0] && LtQ[m,-1/2]
```

2: $\int \mathrm{Sin}[e+f\,x]^2\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\,\bigwedge\,m\nless-\frac{1}{2}$

Derivation: Nondegenerate sine recurrence 1b with $A\to a^2,\ B\to 2\,a\,b,\ C\to b^2,\ m\to 0,\ p\to 0$

Rule: If $a^2-b^2==0\,\bigwedge\,m\nless-\frac{1}{2}$, **then**

$$\int \mathrm{Sin}[e+f\,x]^2\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx\,\longrightarrow$$
$$-\frac{\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}}{b\,f\,(m+2)}+\frac{1}{b\,(m+2)}\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(b\,(m+1)-a\,\mathrm{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[sin[e_.+f_.*x_]^2*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  -Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b*f*(m+2)) +
  1/(b*(m+2))*Int[(a+b*Sin[e+f*x])^m*(b*(m+1)-a*Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,m},x] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1/2]]
```

2. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^2\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0$

   1: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^2\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,m<-1$

Derivation: Singly degenerate sine recurrence 2a with $A\to c,\ B\to d,\ n\to 1,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,m<-1$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^2\,dx\,\longrightarrow$$

$$\frac{(b\,c-a\,d)\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])}{a\,f\,(2\,m+1)}+$$

$$\frac{1}{a\,b\,(2\,m+1)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\left(a\,c\,d\,(m-1)+b\left(d^2+c^2\,(m+1)\right)+d\,(a\,d\,(m-1)+b\,c\,(m+2))\,\mathrm{Sin}[e+f\,x]\right)dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^2,x_Symbol] :=
  (b*c-a*d)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])/(a*f*(2*m+1)) +
  1/(a*b*(2*m+1))*Int[(a+b*Sin[e+f*x])^(m+1)*Simp[a*c*d*(m-1)+b*(d^2+c^2*(m+1))+d*(a*d*(m-1)+b*c*(m+2))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && LtQ[m,-1]
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^2\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2=0\,\wedge\,m\not<-1$

- **Derivation:** Nondegenerate sine recurrence 1b with $A\to a^2,\ B\to 2\,a\,b,\ C\to b^2,\ m\to 0,\ p\to 0$

- **Rule:** If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2=0\,\wedge\,m\not<-1$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^2\,dx\;\longrightarrow$$

$$-\frac{d^2\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}}{b\,f\,(m+2)}+$$

$$\frac{1}{b\,(m+2)}\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,\left(b\,\left(d^2\,(m+1)+c^2\,(m+2)\right)-d\,(a\,d-2\,b\,c\,(m+2))\,\mathrm{Sin}[e+f\,x]\right)dx$$

- **Program code:**

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^2,x_Symbol] :=
  -d^2*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b*f*(m+2)) +
  1/(b*(m+2))*Int[(a+b*Sin[e+f*x])^m*Simp[b*(d^2*(m+1)+c^2*(m+2))-d*(a*d-2*b*c*(m+2))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1]]
```

3. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>1$

1: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>1\,\wedge\,n<-1$

Derivation: Singly degenerate sine recurrence 1a with $A\to a,\ B\to b,\ m\to m-1,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>1\,\wedge\,n<-1$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\ \longrightarrow$$

$$-\frac{b^2\,(b\,c-a\,d)\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m-2}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+1}}{d\,f\,(n+1)\,(b\,c+a\,d)}+$$

$$\frac{b^2}{d\,(n+1)\,(b\,c+a\,d)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m-2}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+1}\,(a\,c\,(m-2)-b\,d\,(m-2\,n-4)-(b\,c\,(m-1)-a\,d\,(m+2\,n+1))\,\mathrm{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -b^2*(b*c-a*d)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m-2)*(c+d*Sin[e+f*x])^(n+1)/(d*f*(n+1)*(b*c+a*d)) +
  b^2/(d*(n+1)*(b*c+a*d))*Int[(a+b*Sin[e+f*x])^(m-2)*(c+d*Sin[e+f*x])^(n+1)*
    Simp[a*c*(m-2)-b*d*(m-2*n-4)-(b*c*(m-1)-a*d*(m+2*n+1))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[m,1] && LtQ[n,-1] &&
  (IntegersQ[2*m,2*n] || IntegerQ[m+1/2] || IntegerQ[m] && EqQ[c,0])
```

2: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>1\,\wedge\,n\nless -1$

Derivation: Singly degenerate sine recurrence 1b with $A\to a,\ B\to b,\ m\to m-1,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>1\,\wedge\,n\nless -1$**, then**

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx\,\longrightarrow$$

$$-\frac{b^2\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m-2}\,(c+d\,\text{Sin}[e+f\,x])^{n+1}}{d\,f\,(m+n)}+$$

$$\frac{1}{d\,(m+n)}\int (a+b\,\text{Sin}[e+f\,x])^{m-2}\,(c+d\,\text{Sin}[e+f\,x])^n\,\cdot$$

$$\left(a\,b\,c\,(m-2)+b^2\,d\,(n+1)+a^2\,d\,(m+n)-b\,(b\,c\,(m-1)-a\,d\,(3\,m+2\,n-2))\,\text{Sin}[e+f\,x]\right)dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -b^2*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m-2)*(c+d*Sin[e+f*x])^(n+1)/(d*f*(m+n)) +
  1/(d*(m+n))*Int[(a+b*Sin[e+f*x])^(m-2)*(c+d*Sin[e+f*x])^n*
     Simp[a*b*c*(m-2)+b^2*d*(n+1)+a^2*d*(m+n)-b*(b*c*(m-1)-a*d*(3*m+2*n-2))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[m,1] && Not[LtQ[n,-1]] &&
  (IntegersQ[2*m,2*n] || IntegerQ[m+1/2] || IntegerQ[m] && EqQ[c,0])
```

4. $\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 == 0 \land c^2 - d^2 \neq 0 \land m < -1$

   1. $\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 == 0 \land c^2 - d^2 \neq 0 \land m < -1 \land n > 0$

      1: $\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 == 0 \land c^2 - d^2 \neq 0 \land m < -1 \land 0 < n < 1$

- **Derivation: Singly degenerate sine recurrence 2a with $A \to 1,\ B \to 0,\ p \to 0$**

- **Rule: If $b c - a d \neq 0 \land a^2 - b^2 == 0 \land c^2 - d^2 \neq 0 \land m < -1 \land 0 < n < 1$, then**

$$\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n \, dx \longrightarrow$$

$$\frac{b \operatorname{Cos}[e + f x] (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n}{a f (2 m + 1)} -$$

$$\frac{1}{a b (2 m + 1)} \int (a + b \operatorname{Sin}[e + f x])^{m+1} (c + d \operatorname{Sin}[e + f x])^{n-1} (a d n - b c (m + 1) - b d (m + n + 1) \operatorname{Sin}[e + f x]) \, dx$$

- **Program code:**

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  b*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n/(a*f*(2*m+1)) -
  1/(a*b*(2*m+1))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^(n-1)*Simp[a*d*n-b*c*(m+1)-b*d*(m+n+1)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[m,-1] && LtQ[0,n,1] &&
  (IntegersQ[2*m,2*n] || IntegerQ[m] && EqQ[c,0])
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m<-1\,\land\,n>1$

Derivation: Singly degenerate sine recurrence 2a with $A\to c,\ B\to d,\ n\to n-1,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m<-1\,\land\,n>1$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\;\longrightarrow$$

$$\frac{(b\,c-a\,d)\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^{n-1}}{a\,f\,(2\,m+1)}+$$

$$\frac{1}{a\,b\,(2\,m+1)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n-2}\left(b\left(c^2\,(m+1)+d^2\,(n-1)\right)+a\,c\,d\,(m-n+1)+d\,(a\,d\,(m-n+1)+b\,c\,(m+n))\,\mathrm{Sin}[e+f\,x]\right)dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  (b*c-a*d)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n-1)/(a*f*(2*m+1)) +
  1/(a*b*(2*m+1))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^(n-2)*
    Simp[b*(c^2*(m+1)+d^2*(n-1))+a*c*d*(m-n+1)+d*(a*d*(m-n+1)+b*c*(m+n))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[m,-1] && GtQ[n,1] &&
  (IntegersQ[2*m,2*n] || IntegerQ[m] && EqQ[c,0])
```

2: $\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 == 0 \land c^2 - d^2 \neq 0 \land m < -1 \land n \ngtr 0$

Derivation: Singly degenerate sine recurrence 2b with $A \to 1, B \to 0, p \to 0$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 == 0 \land c^2 - d^2 \neq 0 \land m < -1 \land n \ngtr 0$**, then**

$$\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n \, dx \longrightarrow$$

$$\frac{b^2 \operatorname{Cos}[e + f x] (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^{n+1}}{a f (2 m + 1) (b c - a d)} +$$

$$\frac{1}{a (2 m + 1) (b c - a d)} \int (a + b \operatorname{Sin}[e + f x])^{m+1} (c + d \operatorname{Sin}[e + f x])^n (b c (m + 1) - a d (2 m + n + 2) + b d (m + n + 2) \operatorname{Sin}[e + f x]) \, dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  b^2*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(a*f*(2*m+1)*(b*c-a*d)) +
  1/(a*(2*m+1)*(b*c-a*d))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n*
    Simp[b*c*(m+1)-a*d*(2*m+n+2)+b*d*(m+n+2)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[m,-1] && Not[GtQ[n,0]] &&
  (IntegersQ[2*m,2*n] || IntegerQ[m] && EqQ[c,0])
```

5. $\int \frac{(c+d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\;\land\;a^2-b^2==0\;\land\;c^2-d^2\neq 0$

1: $\int \frac{(c+d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\;\land\;a^2-b^2==0\;\land\;c^2-d^2\neq 0\;\land\;n>1$

Derivation: Singly degenerate sine recurrence 2a with $A\to c,\;B\to d,\;m\to -1,\;n\to n-1,\;p\to 0$

Rule: If $b\,c-a\,d\neq 0\;\land\;a^2-b^2==0\;\land\;c^2-d^2\neq 0\;\land\;n>1$**, then**

$$\int \frac{(c+d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx\;\longrightarrow$$

$$-\frac{(b\,c-a\,d)\,\mathrm{Cos}[e+f\,x]\,(c+d\,\mathrm{Sin}[e+f\,x])^{n-1}}{a\,f\,(a+b\,\mathrm{Sin}[e+f\,x])}-\frac{d}{a\,b}\int (c+d\,\mathrm{Sin}[e+f\,x])^{n-2}\,(b\,d\,(n-1)-a\,c\,n+(b\,c\,(n-1)-a\,d\,n)\,\mathrm{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(c_.+d_.*sin[e_.+f_.*x_])^n_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -(b*c-a*d)*Cos[e+f*x]*(c+d*Sin[e+f*x])^(n-1)/(a*f*(a+b*Sin[e+f*x])) -
  d/(a*b)*Int[(c+d*Sin[e+f*x])^(n-2)*Simp[b*d*(n-1)-a*c*n+(b*c*(n-1)-a*d*n)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[n,1] && (IntegerQ[2*n] || EqQ[c,0])
```

2: $\int \frac{(c+d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\;\land\;a^2-b^2==0\;\land\;c^2-d^2\neq 0\;\land\;n<0$

Derivation: Singly degenerate sine recurrence 2b with $A\to 1,\;B\to 0,\;m\to -1,\;p\to 0$

Rule: If $b\,c-a\,d\neq 0\;\land\;a^2-b^2==0\;\land\;c^2-d^2\neq 0\;\land\;n<0$**, then**

$$\int \frac{(c+d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx\;\longrightarrow$$

$$-\frac{b^2\,\mathrm{Cos}[e+f\,x]\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+1}}{a\,f\,(b\,c-a\,d)\,(a+b\,\mathrm{Sin}[e+f\,x])}+\frac{d}{a\,(b\,c-a\,d)}\int (c+d\,\mathrm{Sin}[e+f\,x])^n\,(a\,n-b\,(n+1)\,\mathrm{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(c_.+d_.*sin[e_.+f_.*x_])^n_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -b^2*Cos[e+f*x]*(c+d*Sin[e+f*x])^(n+1)/(a*f*(b*c-a*d)*(a+b*Sin[e+f*x])) +
  d/(a*(b*c-a*d))*Int[(c+d*Sin[e+f*x])^n*(a*n-b*(n+1)*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[n,0] && (IntegerQ[2*n] || EqQ[c,0])
```

3: $\int \frac{(c+d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0 \wedge a^2-b^2=0 \wedge c^2-d^2\neq 0$

Derivation: Singly degenerate sine recurrence 2a with $A\to 1,\ B\to 0,\ m\to -1,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0 \wedge a^2-b^2=0 \wedge c^2-d^2\neq 0$, then

$$\int \frac{(c+d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx \longrightarrow$$
$$-\frac{b\,\mathrm{Cos}[e+f\,x]\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{a\,f\,(a+b\,\mathrm{Sin}[e+f\,x])}+\frac{d\,n}{a\,b}\int (c+d\,\mathrm{Sin}[e+f\,x])^{n-1}\,(a-b\,\mathrm{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(c_.+d_.*sin[e_.+f_.*x_])^n_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -b*Cos[e+f*x]*(c+d*Sin[e+f*x])^n/(a*f*(a+b*Sin[e+f*x])) +
  d*n/(a*b)*Int[(c+d*Sin[e+f*x])^(n-1)*(a-b*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && (IntegerQ[2*n] || EqQ[c,0])
```

6. $\int \sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0 \wedge a^2-b^2=0 \wedge c^2-d^2\neq 0$

1. $\int \sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0 \wedge a^2-b^2=0 \wedge c^2-d^2\neq 0 \wedge 2\,n\in\mathbb{Z}$

1: $\int \sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0 \wedge a^2-b^2=0 \wedge c^2-d^2\neq 0 \wedge n>0$

Derivation: Singly degenerate sine recurrence 1b with $A\to c,\ B\to d,\ m\to \frac{1}{2},\ n\to n-1,\ p\to 0$ and algebraic simplification

Rule: If $b\,c-a\,d\neq 0 \wedge a^2-b^2=0 \wedge c^2-d^2\neq 0 \wedge n>0$, then

$$\int \sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx \longrightarrow$$
$$-\frac{2\,b\,\mathrm{Cos}[e+f\,x]\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{f\,(2\,n+1)\,\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}+\frac{2\,n\,(b\,c+a\,d)}{b\,(2\,n+1)}\int \sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n-1}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -2*b*Cos[e+f*x]*(c+d*Sin[e+f*x])^n/(f*(2*n+1)*Sqrt[a+b*Sin[e+f*x]]) +
  2*n*(b*c+a*d)/(b*(2*n+1))*Int[Sqrt[a+b*Sin[e+f*x]]*(c+d*Sin[e+f*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[n,0] && IntegerQ[2*n]
```

2. $\int \sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\land a^2-b^2==0\land c^2-d^2\neq 0\land n<-1$

1: $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{(c+d\,\text{Sin}[e+f\,x])^{3/2}}\,dx$ when $b\,c-a\,d\neq 0\land a^2-b^2==0\land c^2-d^2\neq 0$

- **Derivation: Singly degenerate sine recurrence 1a with** $A\to 1,\ B\to 0,\ m\to \frac{1}{2},\ n\to -\frac{3}{2}\ p\to 0$
- **Derivation: Singly degenerate sine recurrence 1c with** $A\to a,\ B\to b,\ m\to -\frac{1}{2},\ n\to -\frac{3}{2},\ p\to 0$
- **Rule: If** $b\,c-a\,d\neq 0\land a^2-b^2==0\land c^2-d^2\neq 0$**, then**

$$\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{(c+d\,\text{Sin}[e+f\,x])^{3/2}}\,dx\longrightarrow -\frac{2\,b^2\,\text{Cos}[e+f\,x]}{f\,(b\,c+a\,d)\,\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}$$

- **Program code:**

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/(c_.+d_.*sin[e_.+f_.*x_])^(3/2),x_Symbol] :=
  -2*b^2*Cos[e+f*x]/(f*(b*c+a*d)*Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]]) /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2: $\int \sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\land a^2-b^2==0\land c^2-d^2\neq 0\land n<-1$

- **Derivation: Singly degenerate sine recurrence 1c with** $A\to a,\ B\to b,\ m\to -\frac{1}{2},\ p\to 0$ **and algebraic simplification**
- **Rule: If** $b\,c-a\,d\neq 0\land a^2-b^2==0\land c^2-d^2\neq 0\land n<-1$**, then**

$$\int \sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])^n\,dx\longrightarrow$$

$$\frac{(b\,c-a\,d)\,\text{Cos}[e+f\,x]\,(c+d\,\text{Sin}[e+f\,x])^{n+1}}{f\,(n+1)\,\left(c^2-d^2\right)\sqrt{a+b\,\text{Sin}[e+f\,x]}}+\frac{(2\,n+3)\,(b\,c-a\,d)}{2\,b\,(n+1)\,\left(c^2-d^2\right)}\int \sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])^{n+1}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  (b*c-a*d)*Cos[e+f*x]*(c+d*Sin[e+f*x])^(n+1)/(f*(n+1)*(c^2-d^2)*Sqrt[a+b*Sin[e+f*x]]) +
  (2*n+3)*(b*c-a*d)/(2*b*(n+1)*(c^2-d^2))*Int[Sqrt[a+b*Sin[e+f*x]]*(c+d*Sin[e+f*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[n,-1] && NeQ[2*n+3,0] && IntegerQ[2*n]
```

3: $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{c+d\,\text{Sin}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\wedge a^2-b^2==0\wedge c^2-d^2\neq 0$

Author: Martin Welz on 24 June 2011; generalized by Albert Rich 14 April 2014

- **Derivation: Integration by substitution**

- **Basis: If** $a^2-b^2==0$**, then** $\frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{c+d\,\text{Sin}[e+f\,x]}==-\frac{2\,b}{f}\,\text{Subst}\left[\frac{1}{b\,c+a\,d-d\,x^2},\,x,\,\frac{b\,\text{Cos}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}}\right]\partial_x\frac{b\,\text{Cos}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}}$

- **Rule: If** $b\,c-a\,d\neq 0\wedge a^2-b^2==0\wedge c^2-d^2\neq 0$**, then**

$$\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{c+d\,\text{Sin}[e+f\,x]}\,dx\,\longrightarrow\,-\frac{2\,b}{f}\,\text{Subst}\left[\int\frac{1}{b\,c+a\,d-d\,x^2}\,dx,\,x,\,\frac{b\,\text{Cos}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -2*b/f*Subst[Int[1/(b*c+a*d-d*x^2),x],x,b*Cos[e+f*x]/Sqrt[a+b*Sin[e+f*x]]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

4. $\int \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\wedge a^2-b^2==0\wedge c^2-d^2\neq 0$

1: $\int \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}{\sqrt{d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $a^2-b^2==0\wedge d==\frac{a}{b}$

Author: Martin Welz on 24 June 2011

Derivation: Integration by substitution

- **Basis: If** $a^2-b^2==0\wedge d==\frac{a}{b}$, **then** $\frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}{\sqrt{d\,\mathrm{Sin}[e+f\,x]}}==-\frac{2}{f}\,\mathrm{Subst}\left[\frac{1}{\sqrt{1-\frac{x^2}{a}}},\,x,\,\frac{b\,\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\right]\partial_x\frac{b\,\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}$

- **Rule: If** $a^2-b^2==0\wedge d==\frac{a}{b}$, **then**

$$\int \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}{\sqrt{d\,\mathrm{Sin}[e+f\,x]}}\,dx\longrightarrow -\frac{2}{f}\,\mathrm{Subst}\left[\int \frac{1}{\sqrt{1-\frac{x^2}{a}}}\,dx,\,x,\,\frac{b\,\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/Sqrt[d_.*sin[e_.+f_.*x_]],x_Symbol] :=
  -2/f*Subst[Int[1/Sqrt[1-x^2/a],x],x,b*Cos[e+f*x]/Sqrt[a+b*Sin[e+f*x]]] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && EqQ[d,a/b]
```

2: $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\wedge a^2-b^2==0\wedge c^2-d^2\neq 0$

Author: Martin Welz on 10 March 2011

Derivation: Integration by substitution

- **Basis: If** $a^2-b^2==0\wedge c^2-d^2\neq 0$**, then** $\frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{c+d\,\text{Sin}[e+f\,x]}}==-\frac{2\,b}{f}\,\text{Subst}\left[\frac{1}{b+d\,x^2},\,x,\,\frac{b\,\text{Cos}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\right]\partial_x\frac{b\,\text{Cos}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}$

- **Note: The above identity is not valid if** $b\,c-a\,d\neq 0\wedge a^2-b^2==0\wedge c^2-d^2==0$**, since the derivative vanishes!**

Rule: If $b\,c-a\,d\neq 0\wedge a^2-b^2==0\wedge c^2-d^2\neq 0$**, then**

$$\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx\to -\frac{2\,b}{f}\,\text{Subst}\left[\int \frac{1}{b+d\,x^2}\,dx,\,x,\,\frac{b\,\text{Cos}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/Sqrt[c_.+d_.*sin[e_.+f_.*x_]],x_Symbol] :=
  -2*b/f*Subst[Int[1/(b+d*x^2),x],x,b*Cos[e+f*x]/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]])] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2: $\int\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2=0\,\wedge\,c^2-d^2\neq 0\,\wedge\,2\,n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis: If** $a^2-b^2=0$, **then** $\partial_x\frac{\mathrm{Cos}[e+f\,x]}{\sqrt{a-b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}=0$
- **Basis:** $\mathrm{Cos}[e+f\,x]=\frac{1}{f}\,\partial_x\mathrm{Sin}[e+f\,x]$
- **Rule: If** $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2=0\,\wedge\,c^2-d^2\neq 0\,\wedge\,2\,n\notin\mathbb{Z}$, **then**

$$\int\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\;\longrightarrow$$

$$\frac{a^2\,\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sin}[e+f\,x]}}\int\frac{\mathrm{Cos}[e+f\,x]\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{\sqrt{a-b\,\mathrm{Sin}[e+f\,x]}}\,dx\;\longrightarrow$$

$$\frac{a^2\,\mathrm{Cos}[e+f\,x]}{f\,\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sin}[e+f\,x]}}\,\mathrm{Subst}\Big[\int\frac{(c+d\,x)^n}{\sqrt{a-b\,x}}\,dx,\,x,\,\mathrm{Sin}[e+f\,x]\Big]$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  a^2*Cos[e+f*x]/(f*Sqrt[a+b*Sin[e+f*x]]*Sqrt[a-b*Sin[e+f*x]])*Subst[Int[(c+d*x)^n/Sqrt[a-b*x],x],x,Sin[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && Not[IntegerQ[2*n]]
```

7. $\int\frac{(c+d\,\mathrm{Sin}[e+f\,x])^n}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2=0\,\wedge\,c^2-d^2\neq 0$

1. $\int\frac{(c+d\,\mathrm{Sin}[e+f\,x])^n}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2=0\,\wedge\,c^2-d^2\neq 0\,\wedge\,n>0$

1: $\int\frac{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2=0\,\wedge\,c^2-d^2\neq 0$

Derivation: Algebraic expansion

- **Basis:** $\frac{\sqrt{c+d\,z}}{\sqrt{a+b\,z}}=\frac{b\,c-a\,d}{b\,\sqrt{a+b\,z}\,\sqrt{c+d\,z}}+\frac{d\,\sqrt{a+b\,z}}{b\,\sqrt{c+d\,z}}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2=0\,\wedge\,c^2-d^2\neq 0$, **then**

$$\int \frac{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\,dx \to \frac{d}{b}\int \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx + \frac{b\,c-a\,d}{b}\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[Sqrt[c_.+d_.*sin[e_.+f_.*x_]]/Sqrt[a_+b_.*sin[e_.+f_.*x_]],x_Symbol] :=
  d/b*Int[Sqrt[a+b*Sin[e+f*x]]/Sqrt[c+d*Sin[e+f*x]],x] +
  (b*c-a*d)/b*Int[1/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2: $\int \frac{(c+d\,\mathrm{Sin}[e+f\,x])^n}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0 \land a^2-b^2=0 \land c^2-d^2\neq 0 \land n>1$

Derivation: Singly degenerate sine recurrence 2c with $A\to c,\ B\to d,\ m\to \frac{1}{2},\ n\to n-1,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0 \land a^2-b^2=0 \land c^2-d^2\neq 0 \land n>1$, **then**

$$\int \frac{(c+d\,\mathrm{Sin}[e+f\,x])^n}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\,dx \to$$
$$-\frac{2\,d\,\mathrm{Cos}[e+f\,x]\,(c+d\,\mathrm{Sin}[e+f\,x])^{n-1}}{f\,(2\,n-1)\,\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}} -$$
$$\frac{1}{b\,(2\,n-1)}\int \Big((c+d\,\mathrm{Sin}[e+f\,x])^{n-2}\,\big(a\,c\,d-b\,\big(2\,d^2\,(n-1)+c^2\,(2\,n-1)\big)+d\,(a\,d-b\,c\,(4\,n-3))\,\mathrm{Sin}[e+f\,x]\big)\Big)\Big/\Big(\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\Big)\,dx$$

Program code:

```
Int[(c_.+d_.*sin[e_.+f_.*x_])^n_/Sqrt[a_+b_.*sin[e_.+f_.*x_]],x_Symbol] :=
  -2*d*Cos[e+f*x]*(c+d*Sin[e+f*x])^(n-1)/(f*(2*n-1)*Sqrt[a+b*Sin[e+f*x]]) -
  1/(b*(2*n-1))*Int[(c+d*Sin[e+f*x])^(n-2)/Sqrt[a+b*Sin[e+f*x]]*
    Simp[a*c*d-b*(2*d^2*(n-1)+c^2*(2*n-1))+d*(a*d-b*c*(4*n-3))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[n,1] && IntegerQ[2*n]
```

2: $\int \frac{(c+d\,\mathrm{Sin}[e+f\,x])^n}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0 \land a^2-b^2=0 \land c^2-d^2\neq 0 \land n<-1$

Derivation: Singly degenerate sine recurrence 1c with $A\to 1,\ B\to 0,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0 \land a^2-b^2=0 \land c^2-d^2\neq 0 \land n<-1$, **then**

$$\int \frac{(c+d\,\text{Sin}[e+f\,x])^n}{\sqrt{a+b\,\text{Sin}[e+f\,x]}}\,dx \longrightarrow$$
$$-\frac{d\,\text{Cos}[e+f\,x]\,(c+d\,\text{Sin}[e+f\,x])^{n+1}}{f\,(n+1)\,\left(c^2-d^2\right)\sqrt{a+b\,\text{Sin}[e+f\,x]}}-\frac{1}{2\,b\,(n+1)\,\left(c^2-d^2\right)}\int\frac{(c+d\,\text{Sin}[e+f\,x])^{n+1}\,(a\,d-2\,b\,c\,(n+1)+b\,d\,(2\,n+3)\,\text{Sin}[e+f\,x])}{\sqrt{a+b\,\text{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[(c_.+d_.*sin[e_.+f_.*x_])^n_/Sqrt[a_+b_.*sin[e_.+f_.*x_]],x_Symbol] :=
  -d*Cos[e+f*x]*(c+d*Sin[e+f*x])^(n+1)/(f*(n+1)*(c^2-d^2)*Sqrt[a+b*Sin[e+f*x]]) -
  1/(2*b*(n+1)*(c^2-d^2))*Int[(c+d*Sin[e+f*x])^(n+1)*Simp[a*d-2*b*c*(n+1)+b*d*(2*n+3)*Sin[e+f*x],x]/Sqrt[a+b*Sin[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[n,-1] && IntegerQ[2*n]
```

3: $\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0$

Derivation: Algebraic expansion

- **Basis:** $\frac{1}{\sqrt{a+b\,z}\,(c+d\,z)}==\frac{b}{(b\,c-a\,d)\,\sqrt{a+b\,z}}-\frac{d\,\sqrt{a+b\,z}}{(b\,c-a\,d)\,(c+d\,z)}$

- **Rule: If** $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0$**, then**

$$\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx \longrightarrow \frac{b}{b\,c-a\,d}\int\frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}}\,dx-\frac{d}{b\,c-a\,d}\int\frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{c+d\,\text{Sin}[e+f\,x]}\,dx$$

Program code:

```
Int[1/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*(c_.+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  b/(b*c-a*d)*Int[1/Sqrt[a+b*Sin[e+f*x]],x] - d/(b*c-a*d)*Int[Sqrt[a+b*Sin[e+f*x]]/(c+d*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

4. $\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0$

1: $\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $a^2-b^2==0\,\bigwedge\,d==\frac{a}{b}\,\bigwedge\,a>0$

Author: Martin Welz on 24 June 2011

Derivation: Integration by substitution

- **Basis: If** $a^2-b^2==0\,\bigwedge\,d==\frac{a}{b}\,\bigwedge\,a>0$**, then** $\frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{d\,\mathrm{Sin}[e+f\,x]}}==-\frac{\sqrt{2}}{\sqrt{a}\,f}\,\mathrm{Subst}\left[\frac{1}{\sqrt{1-x^2}},\,x,\,\frac{b\,\mathrm{Cos}[e+f\,x]}{a+b\,\mathrm{Sin}[e+f\,x]}\right]\partial_x\frac{b\,\mathrm{Cos}[e+f\,x]}{a+b\,\mathrm{Sin}[e+f\,x]}$

Basis: $F(z\mid 0)==z$

Note: This is a special case of the rule for $a^2\neq b^2$**.**

- **Rule: If** $a^2-b^2==0\,\bigwedge\,d==\frac{a}{b}\,\bigwedge\,a>0$**, then**

$$\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{d\,\mathrm{Sin}[e+f\,x]}}\,dx\,\longrightarrow\,-\frac{\sqrt{2}}{\sqrt{a}\,f}\,\mathrm{Subst}\left[\int\frac{1}{\sqrt{1-x^2}}\,dx,\,x,\,\frac{b\,\mathrm{Cos}[e+f\,x]}{a+b\,\mathrm{Sin}[e+f\,x]}\right]$$

Program code:

```
Int[1/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  -Sqrt[2]/(Sqrt[a]*f)*Subst[Int[1/Sqrt[1-x^2],x],x,b*Cos[e+f*x]/(a+b*Sin[e+f*x])] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && EqQ[d,a/b] && GtQ[a,0]
```

2: $\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0$

Author: Martin Welz on 10 March 2011

Derivation: Integration by substitution

Basis: If $a^2-b^2==0$**, then**

$$\frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}==-\frac{2\,a}{f}\,\mathrm{Subst}\left[\frac{1}{2\,b^2-(a\,c-b\,d)\,x^2},\,x,\,\frac{b\,\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\right]\partial_x\frac{b\,\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}$$

Note: The above identity is not valid if $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2==0$**, since the derivative vanishes!**

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0$**, then**

$$\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\ \sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx \longrightarrow -\frac{2\,a}{f}\,\text{Subst}\Big[\int \frac{1}{2\,b^2-(a\,c-b\,d)\,x^2}\,dx,\ x,\ \frac{b\,\text{Cos}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\ \sqrt{c+d\,\text{Sin}[e+f\,x]}}\Big]$$

Program code:

```
Int[1/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[c_.+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  -2*a/f*Subst[Int[1/(2*b^2-(a*c-b*d)*x^2),x],x,b*Cos[e+f*x]/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]])] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

8: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\ \land\ a^2-b^2==0\ \land\ c^2-d^2\neq 0\ \land\ n>1$

Derivation: Singly degenerate sine recurrence 2c with $A\to c,\ B\to d,\ n\to n-1,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\ \land\ a^2-b^2==0\ \land\ c^2-d^2\neq 0\ \land\ n>1$**, then**

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx \longrightarrow$$

$$-\frac{d\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^{n-1}}{f\,(m+n)}+$$

$$\frac{1}{b\,(m+n)}\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^{n-2}\,\big(d\,(a\,c\,m+b\,d\,(n-1))+b\,c^2\,(m+n)+(d\,(a\,d\,m+b\,c\,(m+2\,n-1)))\,\text{Sin}[e+f\,x]\big)\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -d*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n-1)/(f*(m+n)) +
  1/(b*(m+n))*Int[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n-2)*
    Simp[d*(a*c*m+b*d*(n-1))+b*c^2*(m+n)+d*(a*d*m+b*c*(m+2*n-1))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[n,1] && IntegerQ[n]
```

9. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0$

1: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m\in\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x \frac{\mathrm{Cos}[e+f\,x]}{\sqrt{1+\mathrm{Sin}[e+f\,x]}\,\sqrt{1-\mathrm{Sin}[e+f\,x]}}==0$

- **Basis:** $\mathrm{Cos}[e+f\,x]==\frac{1}{f}\,\partial_x\,\mathrm{Sin}[e+f\,x]$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m\in\mathbb{Z}$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\,\longrightarrow$$

$$\frac{a^m\,\mathrm{Cos}[e+f\,x]}{\sqrt{1+\mathrm{Sin}[e+f\,x]}\,\sqrt{1-\mathrm{Sin}[e+f\,x]}}\int\frac{\mathrm{Cos}[e+f\,x]\,\left(1+\frac{b}{a}\,\mathrm{Sin}[e+f\,x]\right)^{m-\frac{1}{2}}\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{\sqrt{1-\frac{b}{a}\,\mathrm{Sin}[e+f\,x]}}\,dx\,\longrightarrow$$

$$\frac{a^m\,\mathrm{Cos}[e+f\,x]}{f\,\sqrt{1+\mathrm{Sin}[e+f\,x]}\,\sqrt{1-\mathrm{Sin}[e+f\,x]}}\,\mathrm{Subst}\left[\int\frac{\left(1+\frac{b}{a}\,x\right)^{m-\frac{1}{2}}\,(c+d\,x)^n}{\sqrt{1-\frac{b}{a}\,x}}\,dx,\,x,\,\mathrm{Sin}[e+f\,x]\right]$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  a^m*Cos[e+f*x]/(f*Sqrt[1+Sin[e+f*x]]*Sqrt[1-Sin[e+f*x]])*Subst[Int[(1+b/a*x)^(m-1/2)*(c+d*x)^n/Sqrt[1-b/a*x],x],x,Sin[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && IntegerQ[m]
```

2. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m\notin\mathbb{Z}$

1. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(d\,\text{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\land\,m\notin\mathbb{Z}$

1. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(d\,\text{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\land\,m\notin\mathbb{Z}\,\land\,a>0$

1: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(d\,\text{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\bigwedge\,m\notin\mathbb{Z}\,\bigwedge\,a>0\,\bigwedge\,\frac{d}{b}>0$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis: If** $a^2-b^2==0$, **then** $\partial_x\frac{\text{Cos}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{a-b\,\text{Sin}[e+f\,x]}}==0$
- **Basis: If** $a^2-b^2==0$, **then** $\frac{b^2\,\text{Cos}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{a-b\,\text{Sin}[e+f\,x]}}\;\frac{\text{Cos}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{a-b\,\text{Sin}[e+f\,x]}}==1$
- **Basis:** $\frac{\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m-\frac{1}{2}}\,(b\,\text{Sin}[e+f\,x])^n}{\sqrt{a-b\,\text{Sin}[e+f\,x]}}==-\frac{1}{b\,f}\,\text{Subst}\left[\frac{(a-x)^n\,(2\,a-x)^{m-\frac{1}{2}}}{\sqrt{x}},\,x,\,a-b\,\text{Sin}[e+f\,x]\right]\,\partial_x\,(a-b\,\text{Sin}[e+f\,x])$
- **Note: If** $a>0$, **then** $\frac{(a-x)^n\,(2\,a-x)^{m-\frac{1}{2}}}{\sqrt{x}}$ **is integrable in terms of the Appell function without the need for additional piecewise constant extraction.**
- **Rule: If** $a^2-b^2==0\,\bigwedge\,m\notin\mathbb{Z}\,\bigwedge\,a>0\,\bigwedge\,\frac{d}{b}>0$, **then**

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(d\,\text{Sin}[e+f\,x])^n\,dx\;\longrightarrow$$

$$\frac{b^2\left(\frac{d}{b}\right)^n\text{Cos}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{a-b\,\text{Sin}[e+f\,x]}}\int\frac{\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m-\frac{1}{2}}\,(b\,\text{Sin}[e+f\,x])^n}{\sqrt{a-b\,\text{Sin}[e+f\,x]}}\,dx\;\longrightarrow$$

$$-\frac{b\left(\frac{d}{b}\right)^n\text{Cos}[e+f\,x]}{f\,\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{a-b\,\text{Sin}[e+f\,x]}}\,\text{Subst}\left[\int\frac{(a-x)^n\,(2\,a-x)^{m-\frac{1}{2}}}{\sqrt{x}}\,dx,\,x,\,a-b\,\text{Sin}[e+f\,x]\right]$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -b*(d/b)^n*Cos[e+f*x]/(f*Sqrt[a+b*Sin[e+f*x]]*Sqrt[a-b*Sin[e+f*x]])*
    Subst[Int[(a-x)^n*(2*a-x)^(m-1/2)/Sqrt[x],x],x,a-b*Sin[e+f*x]] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[m]] && GtQ[a,0] && GtQ[d/b,0]
```

2: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(d\,\text{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\bigwedge\,m\notin\mathbb{Z}\,\bigwedge\,a>0\,\bigwedge\,\frac{d}{b}\not>0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(d\,\text{Sin}[e+f\,x])^n}{(b\,\text{Sin}[e+f\,x])^n}==0$
- **Rule: If** $a^2-b^2==0\,\bigwedge\,m\notin\mathbb{Z}\,\bigwedge\,a>0\,\bigwedge\,\frac{d}{b}\not>0$, **then**

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(d\,\text{Sin}[e+f\,x])^n\,dx\,\longrightarrow\,\frac{\left(\frac{d}{b}\right)^{\text{IntPart}[n]}\,(d\,\text{Sin}[e+f\,x])^{\text{FracPart}[n]}}{(b\,\text{Sin}[e+f\,x])^{\text{FracPart}[n]}}\int (a+b\,\text{Sin}[e+f\,x])^m\,(b\,\text{Sin}[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  (d/b)^IntPart[n]*(d*Sin[e+f*x])^FracPart[n]/(b*Sin[e+f*x])^FracPart[n]*Int[(a+b*Sin[e+f*x])^m*(b*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[m]] && GtQ[a,0] && Not[GtQ[d/b,0]]
```

2: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(d\,\text{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\wedge\,m\notin\mathbb{Z}\,\wedge\,a\not>0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(a+b\,\text{Sin}[e+f\,x])^m}{\left(1+\frac{b}{a}\,\text{Sin}[e+f\,x]\right)^m}==0$
- **Rule: If** $a^2-b^2==0\,\wedge\,m\notin\mathbb{Z}\,\wedge\,a\not>0$, **then**

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(d\,\text{Sin}[e+f\,x])^n\,dx\,\longrightarrow\,\frac{a^{\text{IntPart}[m]}\,(a+b\,\text{Sin}[e+f\,x])^{\text{FracPart}[m]}}{\left(1+\frac{b}{a}\,\text{Sin}[e+f\,x]\right)^{\text{FracPart}[m]}}\int\left(1+\frac{b}{a}\,\text{Sin}[e+f\,x]\right)^m\,(d\,\text{Sin}[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  a^IntPart[m]*(a+b*Sin[e+f*x])^FracPart[m]/(1+b/a*Sin[e+f*x])^FracPart[m]*
    Int[(1+b/a*Sin[e+f*x])^m*(d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[m]] && Not[GtQ[a,0]]
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis: If** $a^2-b^2==0$**, then** $\partial_x\frac{\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sin}[e+f\,x]}}==0$
- **Basis: If** $a^2-b^2==0$**, then** $\frac{a^2\,\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sin}[e+f\,x]}}\;\frac{\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sin}[e+f\,x]}}==1$
- **Basis:** $\mathrm{Cos}[e+f\,x]==\frac{1}{f}\,\partial_x\mathrm{Sin}[e+f\,x]$
- **Rule: If** $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m\notin\mathbb{Z}$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\;\longrightarrow$$

$$\frac{a^2\,\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sin}[e+f\,x]}}\int\frac{\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m-\frac{1}{2}}\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{\sqrt{a-b\,\mathrm{Sin}[e+f\,x]}}\,dx\;\longrightarrow$$

$$\frac{a^2\,\mathrm{Cos}[e+f\,x]}{f\,\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sin}[e+f\,x]}}\,\mathrm{Subst}\Big[\int\frac{(a+b\,x)^{m-\frac{1}{2}}\,(c+d\,x)^n}{\sqrt{a-b\,x}}\,dx,\,x,\,\mathrm{Sin}[e+f\,x]\Big]$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  a^2*Cos[e+f*x]/(f*Sqrt[a+b*Sin[e+f*x]]*Sqrt[a-b*Sin[e+f*x]])*Subst[Int[(a+b*x)^(m-1/2)*(c+d*x)^n/Sqrt[a-b*x],x],x,Sin[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && Not[IntegerQ[m]]
```

8. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

1. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^2\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0$

1: $\int (b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^2\,dx$

Derivation: Algebraic expansion

Basis: $(c+d\,z)^2 == \frac{2\,c\,d}{b}\,(b\,z)+\left(c^2+d^2\,z^2\right)$

Rule:

$$\int (b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^2\,dx\;\longrightarrow$$
$$\frac{2\,c\,d}{b}\int (b\,\text{Sin}[e+f\,x])^{m+1}\,dx+\int (b\,\text{Sin}[e+f\,x])^m\left(c^2+d^2\,\text{Sin}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^2,x_Symbol] :=
  2*c*d/b*Int[(b*Sin[e+f*x])^(m+1),x] + Int[(b*Sin[e+f*x])^m*(c^2+d^2*Sin[e+f*x]^2),x] /;
FreeQ[{b,c,d,e,f,m},x]
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^2\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m<-1$

Derivation: Nondegenerate sine recurrence 1a with $A\to c^2,\ B\to 2\,c\,d,\ C\to d^2,\ n\to 0,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m<-1$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^2\,dx\,\longrightarrow$$

$$-\frac{\left(b^2\,c^2-2\,a\,b\,c\,d+a^2\,d^2\right)\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}}{b\,f\,(m+1)\,\left(a^2-b^2\right)}-$$

$$\frac{1}{b\,(m+1)\,\left(a^2-b^2\right)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,\left(b\,(m+1)\,\left(2\,b\,c\,d-a\,\left(c^2+d^2\right)\right)+\left(a^2\,d^2-2\,a\,b\,c\,d\,(m+2)+b^2\,\left(d^2\,(m+1)+c^2\,(m+2)\right)\right)\,\mathrm{Sin}[e+f\,x]\right)\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^2,x_Symbol] :=
  -(b^2*c^2-2*a*b*c*d+a^2*d^2)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b*f*(m+1)*(a^2-b^2)) -
  1/(b*(m+1)*(a^2-b^2))*Int[(a+b*Sin[e+f*x])^(m+1)*
    Simp[b*(m+1)*(2*b*c*d-a*(c^2+d^2))+(a^2*d^2-2*a*b*c*d*(m+2)+b^2*(d^2*(m+1)+c^2*(m+2)))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && LtQ[m,-1]
```

3: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^2\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,m\nless -1$

Derivation: Nondegenerate sine recurrence 1b with $A\to a^2,\ B\to 2\,a\,b,\ C\to b^2,\ m\to 0,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,m\nless -1$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^2\,dx\,\longrightarrow$$
$$-\frac{d^2\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{b\,f\,(m+2)}+$$
$$\frac{1}{b\,(m+2)}\int (a+b\,\text{Sin}[e+f\,x])^m\,\left(b\,\left(d^2\,(m+1)+c^2\,(m+2)\right)-d\,(a\,d-2\,b\,c\,(m+2))\,\text{Sin}[e+f\,x]\right)dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^2,x_Symbol] :=
  -d^2*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b*f*(m+2)) +
  1/(b*(m+2))*Int[(a+b*Sin[e+f*x])^m*Simp[b*(d^2*(m+1)+c^2*(m+2))-d*(a*d-2*b*c*(m+2))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && Not[LtQ[m,-1]]
```

x: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(d\,\text{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\,\land\,m\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Note: If terms having the same powers of $\text{Sin}[e+f\,x]$ are collected, this rule results in more compact antiderivatives; however, the number of steps required grows exponentially with m.

Rule: If $a^2-b^2\neq 0\,\land\,m\in\mathbb{Z}^+$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(d\,\text{Sin}[e+f\,x])^n\,dx\,\longrightarrow\,\int \text{ExpandTrig}[(a+b\,\text{Sin}[e+f\,x])^m\,(d\,\text{Sin}[e+f\,x])^n,\,x]\,dx$$

Program code:

```
(* Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[ExpandTrig[(a+b*sin[e+f*x])^m*(d*sin[e+f*x])^n,x],x] /;
FreeQ[{a,b,d,e,f,n},x] && NeQ[a^2-b^2,0] && IGtQ[m,0] *)
```

2. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>2$

1: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>2\,\wedge\,n<-1$

Derivation: Nondegenerate sine recurrence 1a with $A\to c^2,\ B\to 2\,c\,d,\ C\to d^2,\ n\to n-2,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>2\,\wedge\,n<-1$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\ \longrightarrow$$

$$-\left(\left(b^2\,c^2-2\,a\,b\,c\,d+a^2\,d^2\right)\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m-2}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+1}\right)\Big/\left(d\,f\,(n+1)\,\left(c^2-d^2\right)\right)+$$
$$\frac{1}{d\,(n+1)\,\left(c^2-d^2\right)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m-3}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+1}\cdot$$
$$\left(b\,(m-2)\,(b\,c-a\,d)^2+a\,d\,(n+1)\,\left(c\,\left(a^2+b^2\right)-2\,a\,b\,d\right)+\right.$$
$$\left(b\,(n+1)\,\left(a\,b\,c^2+c\,d\,\left(a^2+b^2\right)-3\,a\,b\,d^2\right)-a\,(n+2)\,(b\,c-a\,d)^2\right)\mathrm{Sin}[e+f\,x]+$$
$$\left.b\,\left(b^2\,\left(c^2-d^2\right)-m\,(b\,c-a\,d)^2+d\,n\,\left(2\,a\,b\,c-d\,\left(a^2+b^2\right)\right)\right)\mathrm{Sin}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -(b^2*c^2-2*a*b*c*d+a^2*d^2)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m-2)*(c+d*Sin[e+f*x])^(n+1)/(d*f*(n+1)*(c^2-d^2)) +
  1/(d*(n+1)*(c^2-d^2))*Int[(a+b*Sin[e+f*x])^(m-3)*(c+d*Sin[e+f*x])^(n+1)*
    Simp[b*(m-2)*(b*c-a*d)^2+a*d*(n+1)*(c*(a^2+b^2)-2*a*b*d)+
      (b*(n+1)*(a*b*c^2+c*d*(a^2+b^2)-3*a*b*d^2)-a*(n+2)*(b*c-a*d)^2)*Sin[e+f*x]+
      b*(b^2*(c^2-d^2)-m*(b*c-a*d)^2+d*n*(2*a*b*c-d*(a^2+b^2)))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[m,2] && LtQ[n,-1] && (IntegerQ[m] || IntegersQ[
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>2\,\wedge\,n\nless -1$

Derivation: Nondegenerate sine recurrence 1b with $A\to a^2,\ B\to 2\,a\,b,\ C\to b^2,\ m\to m-2,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>2\,\wedge\,n\nless -1$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\ \longrightarrow$$

$$-\frac{b^2\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m-2}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+1}}{d\,f\,(m+n)}+$$
$$\frac{1}{d\,(m+n)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m-3}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\cdot$$

$$\left(a^3 d (m+n) + b^2 (b c (m-2) + a d (n+1)) - b \left(a b c - b^2 d (m+n-1) - 3 a^2 d (m+n)\right) \text{Sin}[e+f x] - b^2 (b c (m-1) - a d (3 m+2 n-2)) \text{Sin}[e+f x]^2\right) dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -b^2*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m-2)*(c+d*Sin[e+f*x])^(n+1)/(d*f*(m+n)) +
  1/(d*(m+n))*Int[(a+b*Sin[e+f*x])^(m-3)*(c+d*Sin[e+f*x])^n*
    Simp[a^3*d*(m+n)+b^2*(b*c*(m-2)+a*d*(n+1))-
      b*(a*b*c-b^2*d*(m+n-1)-3*a^2*d*(m+n))*Sin[e+f*x]-
      b^2*(b*c*(m-1)-a*d*(3*m+2*n-2))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[m,2] &&
  (IntegerQ[m] || IntegersQ[2*m,2*n]) && Not[IGtQ[n,2] && (Not[IntegerQ[m]] || EqQ[a,0] && NeQ[c,0])]
```

3. $\int (a+b\,\text{Sin}[e+f x])^m (c+d\,\text{Sin}[e+f x])^n\, dx$ when $b c-a d \neq 0 \land a^2-b^2 \neq 0 \land c^2-d^2 \neq 0 \land m < -1$

   1. $\int (a+b\,\text{Sin}[e+f x])^m (c+d\,\text{Sin}[e+f x])^n\, dx$ when $b c-a d \neq 0 \land a^2-b^2 \neq 0 \land c^2-d^2 \neq 0 \land m < -1 \land 0 < n < 2$

      1. $\int (a+b\,\text{Sin}[e+f x])^m (c+d\,\text{Sin}[e+f x])^n\, dx$ when $b c-a d \neq 0 \land a^2-b^2 \neq 0 \land c^2-d^2 \neq 0 \land m < -1 \land 0 < n < 1$

         1. $\int \frac{\sqrt{c+d\,\text{Sin}[e+f x]}}{(a+b\,\text{Sin}[e+f x])^{3/2}}\, dx$ when $b c-a d \neq 0 \land a^2-b^2 \neq 0 \land c^2-d^2 \neq 0$

            1: $\int \frac{\sqrt{d\,\text{Sin}[e+f x]}}{(a+b\,\text{Sin}[e+f x])^{3/2}}\, dx$ when $a^2-b^2 \neq 0$

Derivation: Nondegenerate sine recurrence 1a with $A \to 0,\ B \to d,\ C \to 0,\ m \to -\frac{3}{2},\ n \to -\frac{1}{2},\ p \to 0$

Rule: If $a^2-b^2 \neq 0$**, then**

$$\int \frac{\sqrt{d\,\text{Sin}[e+f x]}}{(a+b\,\text{Sin}[e+f x])^{3/2}}\, dx \longrightarrow -\frac{2 a d\,\text{Cos}[e+f x]}{f \left(a^2-b^2\right) \sqrt{a+b\,\text{Sin}[e+f x]}\ \sqrt{d\,\text{Sin}[e+f x]}} - \frac{d^2}{a^2-b^2} \int \frac{\sqrt{a+b\,\text{Sin}[e+f x]}}{(d\,\text{Sin}[e+f x])^{3/2}}\, dx$$

Program code:

```
Int[Sqrt[d_.*sin[e_.+f_.*x_]]/(a_+b_.*sin[e_.+f_.*x_])^(3/2),x_Symbol] :=
  -2*a*d*Cos[e+f*x]/(f*(a^2-b^2)*Sqrt[a+b*Sin[e+f*x]]*Sqrt[d*Sin[e+f*x]]) -
  d^2/(a^2-b^2)*Int[Sqrt[a+b*Sin[e+f*x]]/(d*Sin[e+f*x])^(3/2),x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0]
```

2: $\int \frac{\sqrt{c+d\,\text{Sin}[e+f\,x]}}{(a+b\,\text{Sin}[e+f\,x])^{3/2}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

Derivation: Algebraic expansion

- **Basis:** $\frac{\sqrt{c+d\,z}}{(a+b\,z)^{3/2}}==\frac{c-d}{a-b}\,\frac{1}{\sqrt{a+b\,z}\,\sqrt{c+d\,z}}-\frac{b\,c-a\,d}{a-b}\,\frac{1+z}{(a+b\,z)^{3/2}\,\sqrt{c+d\,z}}$

- **Rule: If** $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$**, then**

$$\int \frac{\sqrt{c+d\,\text{Sin}[e+f\,x]}}{(a+b\,\text{Sin}[e+f\,x])^{3/2}}\,dx\;\longrightarrow$$

$$\frac{c-d}{a-b}\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx-\frac{b\,c-a\,d}{a-b}\int \frac{1+\text{Sin}[e+f\,x]}{(a+b\,\text{Sin}[e+f\,x])^{3/2}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$$

- **Program code:**

```
Int[Sqrt[c_+d_.*sin[e_.+f_.*x_]]/(a_.+b_.*sin[e_.+f_.*x_])^(3/2),x_Symbol] :=
  (c-d)/(a-b)*Int[1/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]]),x] -
  (b*c-a*d)/(a-b)*Int[(1+Sin[e+f*x])/((a+b*Sin[e+f*x])^(3/2)*Sqrt[c+d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m<-1\,\wedge\,0<n<1$

Derivation: Nondegenerate sine recurrence 1a with $A\to 1,\ B\to 0,\ C\to 0,\ p\to 0$

Derivation: Nondegenerate sine recurrence 1c with $A\to c,\ B\to d,\ C\to 0,\ n\to n-1,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m<-1\,\wedge\,0<n<1$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\,\longrightarrow$$

$$-\frac{b\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{f\,(m+1)\,\left(a^2-b^2\right)}+$$

$$\frac{1}{(m+1)\,\left(a^2-b^2\right)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n-1}\cdot$$

$$\left(a\,c\,(m+1)+b\,d\,n+(a\,d\,(m+1)-b\,c\,(m+2))\,\mathrm{Sin}[e+f\,x]-b\,d\,(m+n+2)\,\mathrm{Sin}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -b*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n/(f*(m+1)*(a^2-b^2)) +
  1/((m+1)*(a^2-b^2))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^(n-1)*
    Simp[a*c*(m+1)+b*d*n+(a*d*(m+1)-b*c*(m+2))*Sin[e+f*x]-b*d*(m+n+2)*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[m,-1] && LtQ[0,n,1] && IntegersQ[2*m,2*n]
```

2. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m<-1\,\wedge\,1<n<2$

   1. $\int \frac{(c+d\,\mathrm{Sin}[e+f\,x])^{3/2}}{(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

      1: $\int \frac{(d\,\mathrm{Sin}[e+f\,x])^{3/2}}{(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}}\,dx$ when $a^2-b^2\neq 0$

Derivation: Algebraic expansion

- **Basis:** $\frac{(d\,z)^{3/2}}{(a+b\,z)^{3/2}}==\frac{d\sqrt{d\,z}}{b\sqrt{a+b\,z}}-\frac{a\,d\sqrt{d\,z}}{b\,(a+b\,z)^{3/2}}$

Rule: If $a^2-b^2\neq 0$**, then**

$$\int \frac{(d\,\mathrm{Sin}[e+f\,x])^{3/2}}{(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}}\,dx\;\longrightarrow\;\frac{d}{b}\int \frac{\sqrt{d\,\mathrm{Sin}[e+f\,x]}}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\,dx-\frac{a\,d}{b}\int \frac{\sqrt{d\,\mathrm{Sin}[e+f\,x]}}{(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}}\,dx$$

Program code:

```
Int[(d_.*sin[e_.+f_.*x_])^(3/2)/(a_+b_.*sin[e_.+f_.*x_])^(3/2),x_Symbol] :=
  d/b*Int[Sqrt[d*Sin[e+f*x]]/Sqrt[a+b*Sin[e+f*x]],x] -
  a*d/b*Int[Sqrt[d*Sin[e+f*x]]/(a+b*Sin[e+f*x])^(3/2),x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0]
```

2: $\int \frac{(c+d\,\mathrm{Sin}[e+f\,x])^{3/2}}{(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

Derivation: Algebraic expansion

- **Basis:** $\frac{(c+d\,z)^{3/2}}{(a+b\,z)^{3/2}} = \frac{d^2\sqrt{a+b\,z}}{b^2\sqrt{c+d\,z}} + \frac{(b\,c-a\,d)\,(b\,c+a\,d+2\,b\,d\,z)}{b^2\,(a+b\,z)^{3/2}\sqrt{c+d\,z}}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$**, then**

$$\int \frac{(c+d\,\mathrm{Sin}[e+f\,x])^{3/2}}{(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}}\,dx \rightarrow \frac{d^2}{b^2}\int \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx + \frac{(b\,c-a\,d)}{b^2}\int \frac{b\,c+a\,d+2\,b\,d\,\mathrm{Sin}[e+f\,x]}{(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[(c_+d_.*sin[e_.+f_.*x_])^(3/2)/(a_.+b_.*sin[e_.+f_.*x_])^(3/2),x_Symbol] :=
  d^2/b^2*Int[Sqrt[a+b*Sin[e+f*x]]/Sqrt[c+d*Sin[e+f*x]],x] +
  (b*c-a*d)/b^2*Int[Simp[b*c+a*d+2*b*d*Sin[e+f*x],x]/((a+b*Sin[e+f*x])^(3/2)*Sqrt[c+d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m<-1\,\wedge\,1<n<2$

Derivation: Nondegenerate sine recurrence 1a with $A\to c,\ B\to d,\ C\to 0,\ n\to n-1,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m<-1\,\wedge\,1<n<2$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\,\longrightarrow$$

$$-\frac{(b\,c-a\,d)\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n-1}}{f\,(m+1)\,\left(a^2-b^2\right)}+$$

$$\frac{1}{(m+1)\,\left(a^2-b^2\right)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n-2}\cdot$$

$$\left(c\,(a\,c-b\,d)\,(m+1)+d\,(b\,c-a\,d)\,(n-1)+(d\,(a\,c-b\,d)\,(m+1)-c\,(b\,c-a\,d)\,(m+2))\,\mathrm{Sin}[e+f\,x]-d\,(b\,c-a\,d)\,(m+n+1)\,\mathrm{Sin}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -(b*c-a*d)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^(n-1)/(f*(m+1)*(a^2-b^2)) +
  1/((m+1)*(a^2-b^2))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^(n-2)*
    Simp[c*(a*c-b*d)*(m+1)+d*(b*c-a*d)*(n-1)+(d*(a*c-b*d)*(m+1)-c*(b*c-a*d)*(m+2))*Sin[e+f*x]-d*(b*c-a*d)*(m+n+1)*Sin[e+f*x]^2,x],x
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[m,-1] && LtQ[1,n,2] && IntegersQ[2*m,2*n]
```

2. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,m<-1\,\land\,n\not>0$

1. $\int \frac{1}{(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$

1: $\int \frac{1}{(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}\,\sqrt{d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0$

Derivation: Nondegenerate sine recurrence 1a with $c\to 0,\ A\to 1,\ B\to 0,\ C\to 0,\ p\to 0,\ m\to -\frac{3}{2},\ n\to -\frac{1}{2}$

Rule: If $a^2-b^2\neq 0$**, then**

$$\int \frac{1}{(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}\,\sqrt{d\,\mathrm{Sin}[e+f\,x]}}\,dx\,\longrightarrow\,\frac{2\,b\,\mathrm{Cos}[e+f\,x]}{f\,\left(a^2-b^2\right)\,\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{d\,\mathrm{Sin}[e+f\,x]}}+\frac{d}{a^2-b^2}\int \frac{b+a\,\mathrm{Sin}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,(d\,\mathrm{Sin}[e+f\,x])^{3/2}}\,dx$$

Program code:

```
Int[1/((a_+b_.*sin[e_.+f_.*x_])^(3/2)*Sqrt[d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  2*b*Cos[e+f*x]/(f*(a^2-b^2)*Sqrt[a+b*Sin[e+f*x]]*Sqrt[d*Sin[e+f*x]]) +
  d/(a^2-b^2)*Int[(b+a*Sin[e+f*x])/(Sqrt[a+b*Sin[e+f*x]]*(d*Sin[e+f*x])^(3/2)),x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0]
```

2: $\int \frac{1}{(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{(a+b\,z)^{3/2}}==\frac{1}{(a-b)\,\sqrt{a+b\,z}}-\frac{b\,(1+z)}{(a-b)\,(a+b\,z)^{3/2}}$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$**, then**

$$\int \frac{1}{(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx\,\longrightarrow$$

$$\frac{1}{a-b}\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx-\frac{b}{a-b}\int \frac{1+\mathrm{Sin}[e+f\,x]}{(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[1/((a_.+b_.*sin[e_.+f_.*x_])^(3/2)*Sqrt[c_.+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  1/(a-b)*Int[1/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]]),x] -
  b/(a-b)*Int[(1+Sin[e+f*x])/((a+b*Sin[e+f*x])^(3/2)*Sqrt[c+d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m<-1\,\wedge\,n\not>0$

Derivation: Nondegenerate sine recurrence 1c with $A\to 1,\ B\to 0,\ C\to 0,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m<-1\,\wedge\,n\not>0$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\;\longrightarrow$$

$$-\frac{b^2\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+1}}{f\,(m+1)\,(b\,c-a\,d)\,\left(a^2-b^2\right)}+$$

$$\frac{1}{(m+1)\,(b\,c-a\,d)\,\left(a^2-b^2\right)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,\cdot$$

$$\left(a\,(b\,c-a\,d)\,(m+1)+b^2\,d\,(m+n+2)-\left(b^2\,c+b\,(b\,c-a\,d)\,(m+1)\right)\mathrm{Sin}[e+f\,x]-b^2\,d\,(m+n+3)\,\mathrm{Sin}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -b^2*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^(n+1)/(f*(m+1)*(b*c-a*d)*(a^2-b^2)) +
  1/((m+1)*(b*c-a*d)*(a^2-b^2))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n*
    Simp[a*(b*c-a*d)*(m+1)+b^2*d*(m+n+2)-(b^2*c+b*(b*c-a*d)*(m+1))*Sin[e+f*x]-b^2*d*(m+n+3)*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[m,-1] && IntegersQ[2*m,2*n] &&
  (EqQ[a,0] && IntegerQ[m] && Not[IntegerQ[n]] || Not[IntegerQ[2*n] && LtQ[n,-1] && (IntegerQ[n] && Not[IntegerQ[m]] || EqQ[a,0])])
```

4: $\int \frac{\sqrt{c+d\,\text{Sin}[e+f\,x]}}{a+b\,\text{Sin}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{c+d\,z}}{a+b\,z}==\frac{d}{b\sqrt{c+d\,z}}+\frac{b\,c-a\,d}{b\,(a+b\,z)\sqrt{c+d\,z}}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$**, then**

$$\int \frac{\sqrt{c+d\,\text{Sin}[e+f\,x]}}{a+b\,\text{Sin}[e+f\,x]}\,dx\,\longrightarrow\,\frac{d}{b}\int \frac{1}{\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx+\frac{b\,c-a\,d}{b}\int \frac{1}{(a+b\,\text{Sin}[e+f\,x])\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[Sqrt[c_.+d_.*sin[e_.+f_.*x_]]/(a_.+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  d/b*Int[1/Sqrt[c+d*Sin[e+f*x]],x] +
  (b*c-a*d)/b*Int[1/((a+b*Sin[e+f*x])*Sqrt[c+d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

5: $\int \frac{(a+b\,\text{Sin}[e+f\,x])^{3/2}}{c+d\,\text{Sin}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{a+b\,z}{c+d\,z}==\frac{b}{d}-\frac{b\,c-a\,d}{d\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$**, then**

$$\int \frac{(a+b\,\text{Sin}[e+f\,x])^{3/2}}{c+d\,\text{Sin}[e+f\,x]}\,dx\,\longrightarrow\,\frac{b}{d}\int \sqrt{a+b\,\text{Sin}[e+f\,x]}\,dx-\frac{b\,c-a\,d}{d}\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{c+d\,\text{Sin}[e+f\,x]}\,dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^(3/2)/(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  b/d*Int[Sqrt[a+b*Sin[e+f*x]],x] - (b*c-a*d)/d*Int[Sqrt[a+b*Sin[e+f*x]]/(c+d*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

6. $\int \frac{1}{(a+b\,\text{Sin}[e+f\,x])\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

1: $\int \frac{1}{(a+b\,\text{Sin}[e+f\,x])\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,c+d>0$

Derivation: Primitive rule

- **Basis: If** $c+d>0$**, then** $\partial_x\,\text{EllipticPi}\left[\frac{2b}{a+b},\ \frac{1}{2}\left(x-\frac{\pi}{2}\right),\ \frac{2d}{c+d}\right]=\frac{(a+b)\sqrt{c+d}}{2\,(a+b\,\text{Sin}[x])\sqrt{c+d\,\text{Sin}[x]}}$

- **Rule: If** $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,c+d>0$**, then**

$$\int \frac{1}{(a+b\,\text{Sin}[e+f\,x])\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx\ \longrightarrow\ \frac{2}{f\,(a+b)\sqrt{c+d}}\,\text{EllipticPi}\left[\frac{2b}{a+b},\ \frac{1}{2}\left(e-\frac{\pi}{2}+f\,x\right),\ \frac{2d}{c+d}\right]$$

- **Program code:**

```
Int[1/((a_.+b_.*sin[e_.+f_.*x_])*Sqrt[c_.+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  2/(f*(a+b)*Sqrt[c+d])*EllipticPi[2*b/(a+b),1/2*(e-Pi/2+f*x),2*d/(c+d)] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[c+d,0]
```

2: $\int \frac{1}{(a+b\,\text{Sin}[e+f\,x])\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,c-d>0$

- **Derivation: Primitive rule**

- **Basis: If** $c-d>0$**, then** $\partial_x\,\text{EllipticPi}\left[-\frac{2b}{a-b},\ \frac{1}{2}\left(x+\frac{\pi}{2}\right),\ -\frac{2d}{c-d}\right]=\frac{(a-b)\sqrt{c-d}}{2\,(a+b\,\text{Sin}[x])\sqrt{c+d\,\text{Sin}[x]}}$

- **Rule: If** $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,c-d>0$**, then**

$$\int \frac{1}{(a+b\,\text{Sin}[e+f\,x])\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx\ \longrightarrow\ \frac{2}{f\,(a-b)\sqrt{c-d}}\,\text{EllipticPi}\left[-\frac{2b}{a-b},\ \frac{1}{2}\left(e+\frac{\pi}{2}+f\,x\right),\ -\frac{2d}{c-d}\right]$$

Program code:

```
Int[1/((a_.+b_.*sin[e_.+f_.*x_])*Sqrt[c_.+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  2/(f*(a-b)*Sqrt[c-d])*EllipticPi[-2*b/(a-b),1/2*(e+Pi/2+f*x),-2*d/(c-d)] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[c-d,0]
```

3: $\int \frac{1}{(a+b\,\mathrm{Sin}[e+f\,x])\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,c+d\ngtr 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{\frac{c+d\,F[x]}{c+d}}}{\sqrt{c+d\,F[x]}} == 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,c+d\ngtr 0$**, then**

$$\int \frac{1}{(a+b\,\mathrm{Sin}[e+f\,x])\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx\;\longrightarrow\;\frac{\sqrt{\frac{c+d\,\mathrm{Sin}[e+f\,x]}{c+d}}}{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\int \frac{1}{(a+b\,\mathrm{Sin}[e+f\,x])\,\sqrt{\frac{c}{c+d}+\frac{d}{c+d}\,\mathrm{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[1/((a_.+b_.*sin[e_.+f_.*x_])*Sqrt[c_.+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  Sqrt[(c+d*Sin[e+f*x])/(c+d)]/Sqrt[c+d*Sin[e+f*x]]*Int[1/((a+b*Sin[e+f*x])*Sqrt[c/(c+d)+d/(c+d)*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && Not[GtQ[c+d,0]]
```

7. $\int \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

1. $\int \frac{\sqrt{b\,\mathrm{Sin}[e+f\,x]}}{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $c^2-d^2\neq 0$

1. $\int \frac{\sqrt{b\,\mathrm{Sin}[e+f\,x]}}{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $c^2-d^2\neq 0\,\wedge\,\frac{c+d}{b}>0$

1: $\int \frac{\sqrt{b\,\mathrm{Sin}[e+f\,x]}}{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $c^2-d^2>0\,\wedge\,\frac{c+d}{b}>0\,\wedge\,c^2>0$

- **Rule: If** $c^2-d^2>0\,\wedge\,\frac{c+d}{b}>0\,\wedge\,c^2>0$**, then**

$$\int \frac{\sqrt{b\,\mathrm{Sin}[e+f\,x]}}{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx\;\longrightarrow$$

$$\frac{2\,c\,\sqrt{b\,(c+d)}\;\text{Tan}[e+f\,x]\,\sqrt{1+\text{Csc}[e+f\,x]}\,\sqrt{1-\text{Csc}[e+f\,x]}}{d\,f\,\sqrt{c^2-d^2}}\,\text{EllipticPi}\left[\frac{c+d}{d},\,\text{ArcSin}\left[\frac{\sqrt{c+d\,\text{Sin}[e+f\,x]}}{\sqrt{b\,\text{Sin}[e+f\,x]}}\Big/\sqrt{\frac{c+d}{b}}\right],\,-\frac{c+d}{c-d}\right]$$

Program code:

```
Int[Sqrt[b_.*sin[e_.+f_.*x_]]/Sqrt[c_+d_.*sin[e_.+f_.*x_]],x_Symbol] :=
  2*c*Rt[b*(c+d),2]*Tan[e+f*x]*Sqrt[1+Csc[e+f*x]]*Sqrt[1-Csc[e+f*x]]/(d*f*Sqrt[c^2-d^2])*
    EllipticPi[(c+d)/d,ArcSin[Sqrt[c+d*Sin[e+f*x]]/Sqrt[b*Sin[e+f*x]]/Rt[(c+d)/b,2]],-(c+d)/(c-d)] /;
FreeQ[{b,c,d,e,f},x] && GtQ[c^2-d^2,0] && PosQ[(c+d)/b] && GtQ[c^2,0]
```

2: $\int \frac{\sqrt{b\,\text{Sin}[e+f\,x]}}{\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $c^2-d^2\neq 0\,\bigwedge\,\frac{c+d}{b}>0$

- **Rule: If** $c^2-d^2\neq 0\,\bigwedge\,\frac{c+d}{b}>0$**, then**

$$\int \frac{\sqrt{b\,\text{Sin}[e+f\,x]}}{\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx\;\longrightarrow$$

$$\frac{2\,b\,\text{Tan}[e+f\,x]}{d\,f}\sqrt{\frac{c+d}{b}}\sqrt{\frac{c\,(1+\text{Csc}[e+f\,x])}{c-d}}\sqrt{\frac{c\,(1-\text{Csc}[e+f\,x])}{c+d}}\,\text{EllipticPi}\left[\frac{c+d}{d},\,\text{ArcSin}\left[\frac{\sqrt{c+d\,\text{Sin}[e+f\,x]}}{\sqrt{b\,\text{Sin}[e+f\,x]}}\Big/\sqrt{\frac{c+d}{b}}\right],\,-\frac{c+d}{c-d}\right]$$

Program code:

```
Int[Sqrt[b_.*sin[e_.+f_.*x_]]/Sqrt[c_+d_.*sin[e_.+f_.*x_]],x_Symbol] :=
  2*b*Tan[e+f*x]/(d*f)*Rt[(c+d)/b,2]*Sqrt[c*(1+Csc[e+f*x])/(c-d)]*Sqrt[c*(1-Csc[e+f*x])/(c+d)]*
    EllipticPi[(c+d)/d,ArcSin[Sqrt[c+d*Sin[e+f*x]]/Sqrt[b*Sin[e+f*x]]/Rt[(c+d)/b,2]],-(c+d)/(c-d)] /;
FreeQ[{b,c,d,e,f},x] && NeQ[c^2-d^2,0] && PosQ[(c+d)/b]
```

2: $\int \frac{\sqrt{b\,\text{Sin}[e+f\,x]}}{\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $c^2-d^2\neq 0\,\bigwedge\,\frac{c+d}{b}\not>0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{F[x]}}{\sqrt{-F[x]}} == 0$

- **Rule: If** $c^2-d^2\neq 0\,\bigwedge\,\frac{c+d}{b}\not>0$**, then**

$$\int \frac{\sqrt{b \operatorname{Sin}[e+f x]}}{\sqrt{c+d \operatorname{Sin}[e+f x]}} \, dx \to \frac{\sqrt{b \operatorname{Sin}[e+f x]}}{\sqrt{-b \operatorname{Sin}[e+f x]}} \int \frac{\sqrt{-b \operatorname{Sin}[e+f x]}}{\sqrt{c+d \operatorname{Sin}[e+f x]}} \, dx$$

Program code:

```
Int[Sqrt[b_.*sin[e_.+f_.*x_]]/Sqrt[c_+d_.*sin[e_.+f_.*x_]],x_Symbol] :=
  Sqrt[b*Sin[e+f*x]]/Sqrt[-b*Sin[e+f*x]]*Int[Sqrt[-b*Sin[e+f*x]]/Sqrt[c+d*Sin[e+f*x]],x] /;
FreeQ[{b,c,d,e,f},x] && NeQ[c^2-d^2,0] && NegQ[(c+d)/b]
```

2. $\int \frac{\sqrt{a+b \operatorname{Sin}[e+f x]}}{\sqrt{c+d \operatorname{Sin}[e+f x]}} \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0$

x: $\int \frac{\sqrt{a+b \operatorname{Sin}[e+f x]}}{\sqrt{d \operatorname{Sin}[e+f x]}} \, dx$ when $a^2 - b^2 \neq 0$

Derivation: Algebraic expansion

- **Basis:** $\frac{\sqrt{a+b z}}{\sqrt{d z}} = \frac{a}{\sqrt{a+b z} \sqrt{d z}} + \frac{b \sqrt{d z}}{d \sqrt{a+b z}}$

- **Rule: If** $a^2 - b^2 \neq 0$**, then**

$$\int \frac{\sqrt{a+b \operatorname{Sin}[e+f x]}}{\sqrt{d \operatorname{Sin}[e+f x]}} \, dx \to a \int \frac{1}{\sqrt{a+b \operatorname{Sin}[e+f x]} \sqrt{d \operatorname{Sin}[e+f x]}} \, dx + \frac{b}{d} \int \frac{\sqrt{d \operatorname{Sin}[e+f x]}}{\sqrt{a+b \operatorname{Sin}[e+f x]}} \, dx$$

Program code:

```
(* Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/Sqrt[d_.*sin[e_.+f_.*x_]],x_Symbol] :=
  a*Int[1/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[d*Sin[e+f*x]]),x] +
  b/d*Int[Sqrt[d*Sin[e+f*x]]/Sqrt[a+b*Sin[e+f*x]],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] *)
```

x: $\int \frac{\sqrt{a+b \operatorname{Sin}[e+f x]}}{\sqrt{d \operatorname{Sin}[e+f x]}} \, dx$ when $a^2 - b^2 \neq 0 \bigwedge \frac{a+b}{d} > 0$

- **Rule: If** $a^2 - b^2 \neq 0 \bigwedge \frac{a+b}{d} > 0$**, then**

$$\int \frac{\sqrt{a+b \operatorname{Sin}[e+f x]}}{\sqrt{d \operatorname{Sin}[e+f x]}} \, dx \to$$

$$\frac{2\,(a+b\,\text{Sin}[e+f\,x])}{d\,f\,\sqrt{\frac{a+b}{d}}\,\text{Cos}[e+f\,x]}\sqrt{\frac{a\,(1-\text{Sin}[e+f\,x])}{a+b\,\text{Sin}[e+f\,x]}}\sqrt{\frac{a\,(1+\text{Sin}[e+f\,x])}{a+b\,\text{Sin}[e+f\,x]}}\,\text{EllipticPi}\Big[\frac{b}{a+b},\,\text{ArcSin}\Big[\sqrt{\frac{a+b}{d}}\,\frac{\sqrt{d\,\text{Sin}[e+f\,x]}}{\sqrt{a+b\,\text{Sin}[e+f\,x]}}\Big],\,-\frac{a-b}{a+b}\Big]$$

Program code:

```
(* Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/Sqrt[d_.*sin[e_.+f_.*x_]],x_Symbol] :=
  2*(a+b*Sin[e+f*x])/(d*f*Rt[(a+b)/d,2]*Cos[e+f*x])*Sqrt[a*(1-Sin[e+f*x])/(a+b*Sin[e+f*x])]*Sqrt[a*(1+Sin[e+f*x])/(a+b*Sin[e+f*x])]
    EllipticPi[b/(a+b),ArcSin[Rt[(a+b)/d,2]*(Sqrt[d*Sin[e+f*x]]/Sqrt[a+b*Sin[e+f*x]])],-(a-b)/(a+b)] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && PosQ[(a+b)/d] *)
```

1: $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\bigwedge\,a^2-b^2\neq 0\,\bigwedge\,c^2-d^2\neq 0\,\bigwedge\,\frac{a+b}{c+d}>0$

- **Rule: If** $b\,c-a\,d\neq 0\,\bigwedge\,a^2-b^2\neq 0\,\bigwedge\,c^2-d^2\neq 0\,\bigwedge\,\frac{a+b}{c+d}>0$**, then**

$$\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx \to$$

$$\frac{2\,(a+b\,\text{Sin}[e+f\,x])}{d\,f\,\sqrt{\frac{a+b}{c+d}}\,\text{Cos}[e+f\,x]}\sqrt{\frac{(b\,c-a\,d)\,(1+\text{Sin}[e+f\,x])}{(c-d)\,(a+b\,\text{Sin}[e+f\,x])}}$$

$$\sqrt{-\frac{(b\,c-a\,d)\,(1-\text{Sin}[e+f\,x])}{(c+d)\,(a+b\,\text{Sin}[e+f\,x])}}\,\text{EllipticPi}\Big[\frac{b\,(c+d)}{d\,(a+b)},\,\text{ArcSin}\Big[\sqrt{\frac{a+b}{c+d}}\,\frac{\sqrt{c+d\,\text{Sin}[e+f\,x]}}{\sqrt{a+b\,\text{Sin}[e+f\,x]}}\Big],\,\frac{(a-b)\,(c+d)}{(a+b)\,(c-d)}\Big]$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/Sqrt[c_.+d_.*sin[e_.+f_.*x_]],x_Symbol] :=
  2*(a+b*Sin[e+f*x])/(d*f*Rt[(a+b)/(c+d),2]*Cos[e+f*x])*
    Sqrt[(b*c-a*d)*(1+Sin[e+f*x])/((c-d)*(a+b*Sin[e+f*x]))]*
    Sqrt[-(b*c-a*d)*(1-Sin[e+f*x])/((c+d)*(a+b*Sin[e+f*x]))]*
    EllipticPi[b*(c+d)/(d*(a+b)),ArcSin[Rt[(a+b)/(c+d),2]*Sqrt[c+d*Sin[e+f*x]]/Sqrt[a+b*Sin[e+f*x]]],(a-b)*(c+d)/((a+b)*(c-d))] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && PosQ[(a+b)/(c+d)]
```

2: $\int \frac{\sqrt{a + b \,\text{Sin}[e + f\,x]}}{\sqrt{c + d\,\text{Sin}[e + f\,x]}}\,dx$ when $b\,c - a\,d \neq 0 \,\bigwedge\, a^2 - b^2 \neq 0 \,\bigwedge\, c^2 - d^2 \neq 0 \,\bigwedge\, \frac{a+b}{c+d} \ngtr 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{F[x]}}{\sqrt{-F[x]}} == 0$

- **Rule: If** $b\,c - a\,d \neq 0 \,\bigwedge\, a^2 - b^2 \neq 0 \,\bigwedge\, c^2 - d^2 \neq 0 \,\bigwedge\, \frac{a+b}{c+d} \ngtr 0$, **then**

$$\int \frac{\sqrt{a + b\,\text{Sin}[e + f\,x]}}{\sqrt{c + d\,\text{Sin}[e + f\,x]}}\,dx \to \frac{\sqrt{-c - d\,\text{Sin}[e + f\,x]}}{\sqrt{c + d\,\text{Sin}[e + f\,x]}} \int \frac{\sqrt{a + b\,\text{Sin}[e + f\,x]}}{\sqrt{-c - d\,\text{Sin}[e + f\,x]}}\,dx$$

- **Program code:**

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/Sqrt[c_.+d_.*sin[e_.+f_.*x_]],x_Symbol] :=
  Sqrt[-c-d*Sin[e+f*x]]/Sqrt[c+d*Sin[e+f*x]]*Int[Sqrt[a+b*Sin[e+f*x]]/Sqrt[-c-d*Sin[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && NegQ[(a+b)/(c+d)]
```

8. $\int \frac{1}{\sqrt{a + b\,\text{Sin}[e + f\,x]}\,\sqrt{c + d\,\text{Sin}[e + f\,x]}}\,dx$ when $b\,c - a\,d \neq 0 \,\wedge\, a^2 - b^2 \neq 0 \,\wedge\, c^2 - d^2 \neq 0$

1. $\int \frac{1}{\sqrt{a + b\,\text{Sin}[e + f\,x]}\,\sqrt{d\,\text{Sin}[e + f\,x]}}\,dx$ when $a^2 - b^2 \neq 0$

1. $\int \frac{1}{\sqrt{a + b\,\text{Sin}[e + f\,x]}\,\sqrt{d\,\text{Sin}[e + f\,x]}}\,dx$ when $a^2 - b^2 < 0 \,\bigwedge\, b^2 > 0$

1: $\int \frac{1}{\sqrt{a + b\,\text{Sin}[e + f\,x]}\,\sqrt{d\,\text{Sin}[e + f\,x]}}\,dx$ when $a^2 - b^2 < 0 \,\bigwedge\, d^2 == 1 \,\bigwedge\, b\,d > 0$

- **Derivation: Integration by substitution**

- **Basis: If** $a^2 - b^2 < 0 \,\bigwedge\, d^2 == 1 \,\bigwedge\, b\,d > 0$, **then**
$\frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{d\,\text{Sin}[e+f\,x]}} == -\frac{2\,d}{f\,\sqrt{a+b\,d}}\,\text{Subst}\left[\frac{1}{\sqrt{1-x^2}\,\sqrt{1+\frac{(a-b\,d)\,x^2}{a+b\,d}}},\,x,\,\frac{\text{Cos}[e+f\,x]}{1+d\,\text{Sin}[e+f\,x]}\right]\,\partial_x \frac{\text{Cos}[e+f\,x]}{1+d\,\text{Sin}[e+f\,x]}$

- **Rule: If** $a^2 - b^2 < 0 \,\bigwedge\, d^2 == 1 \,\bigwedge\, b\,d > 0$, **then**

$$\int \frac{1}{\sqrt{a + b\,\text{Sin}[e + f\,x]}\,\sqrt{d\,\text{Sin}[e + f\,x]}}\,dx \to -\frac{2\,d}{f\,\sqrt{a + b\,d}}\,\text{Subst}\left[\int \frac{1}{\sqrt{1 - x^2}\,\sqrt{1 + \frac{(a-b\,d)\,x^2}{a+b\,d}}}\,dx,\,x,\,\frac{\text{Cos}[e + f\,x]}{1 + d\,\text{Sin}[e + f\,x]}\right]$$

$$\longrightarrow -\frac{2\,d}{f\,\sqrt{a+b\,d}}\,\text{EllipticF}\left[\text{ArcSin}\left[\frac{\text{Cos}[e+f\,x]}{1+d\,\text{Sin}[e+f\,x]}\right],\,-\frac{a-b\,d}{a+b\,d}\right]$$

Program code:

```
Int[1/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  -2*d/(f*Sqrt[a+b*d])*EllipticF[ArcSin[Cos[e+f*x]/(1+d*Sin[e+f*x])],-(a-b*d)/(a+b*d)] /;
FreeQ[{a,b,d,e,f},x] && LtQ[a^2-b^2,0] && EqQ[d^2,1] && GtQ[b*d,0]
```

2: $\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx$ when $a^2-b^2<0\,\land\,b^2>0\,\land\,\neg\left(d^2==1\,\land\,b\,d>0\right)$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{b\,F[x]}}{\sqrt{d\,F[x]}} == 0$
- **Rule:** If $a^2-b^2<0\,\land\,b^2>0\,\land\,\neg\left(d^2==1\,\land\,b\,d>0\right)$, then

$$\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx \longrightarrow \frac{\sqrt{\text{Sign}[b]\,\text{Sin}[e+f\,x]}}{\sqrt{d\,\text{Sin}[e+f\,x]}}\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{\text{Sign}[b]\,\text{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[1/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  Sqrt[Sign[b]*Sin[e+f*x]]/Sqrt[d*Sin[e+f*x]]*Int[1/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[Sign[b]*Sin[e+f*x]]),x] /;
FreeQ[{a,b,d,e,f},x] && LtQ[a^2-b^2,0] && GtQ[b^2,0] && Not[EqQ[d^2,1] && GtQ[b*d,0]]
```

2. $\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0\,\bigwedge\,\frac{a+b}{d}>0$

1: $\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx$ when $a^2-b^2>0\,\bigwedge\,\frac{a+b}{d}>0\,\bigwedge\,a^2>0$

- **Rule: If** $a^2-b^2>0\,\bigwedge\,\frac{a+b}{d}>0\,\bigwedge\,a^2>0$**, then**

$$\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx \to -\frac{2\sqrt{a^2}\,\sqrt{-\text{Cot}[e+f\,x]^2}}{a\,f\,\sqrt{a^2-b^2}\,\text{Cot}[e+f\,x]}\sqrt{\frac{a+b}{d}}\,\text{EllipticF}\Big[\text{ArcSin}\Big[\frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{d\,\text{Sin}[e+f\,x]}}\Big/\sqrt{\frac{a+b}{d}}\Big],-\frac{a+b}{a-b}\Big]$$

Program code:

```
Int[1/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  -2*Sqrt[a^2]*Sqrt[-Cot[e+f*x]^2]/(a*f*Sqrt[a^2-b^2]*Cot[e+f*x])*Rt[(a+b)/d,2]*
    EllipticF[ArcSin[Sqrt[a+b*Sin[e+f*x]]/Sqrt[d*Sin[e+f*x]]/Rt[(a+b)/d,2]],-(a+b)/(a-b)] /;
FreeQ[{a,b,d,e,f},x] && GtQ[a^2-b^2,0] && PosQ[(a+b)/d] && GtQ[a^2,0]
```

2: $\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0\,\bigwedge\,\frac{a+b}{d}>0$

- **Rule: If** $a^2-b^2\neq 0\,\bigwedge\,\frac{a+b}{d}>0$**, then**

$$\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx \to$$

$$-\frac{2\,\text{Tan}[e+f\,x]}{a\,f}\sqrt{\frac{a+b}{d}}\,\sqrt{\frac{a\,(1-\text{Csc}[e+f\,x])}{a+b}}\,\sqrt{\frac{a\,(1+\text{Csc}[e+f\,x])}{a-b}}\,\text{EllipticF}\Big[\text{ArcSin}\Big[\frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{d\,\text{Sin}[e+f\,x]}}\Big/\sqrt{\frac{a+b}{d}}\Big],-\frac{a+b}{a-b}\Big]$$

Program code:

```
Int[1/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  -2*Tan[e+f*x]/(a*f)*Rt[(a+b)/d,2]*Sqrt[a*(1-Csc[e+f*x])/(a+b)]*Sqrt[a*(1+Csc[e+f*x])/(a-b)]*
    EllipticF[ArcSin[Sqrt[a+b*Sin[e+f*x]]/Sqrt[d*Sin[e+f*x]]/Rt[(a+b)/d,2]],-(a+b)/(a-b)] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && PosQ[(a+b)/d]
```

3: $\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0\,\bigwedge\,\frac{a+b}{d}\ngtr 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{-F[x]}}{\sqrt{F[x]}} == 0$

- **Rule: If** $a^2-b^2\neq 0\,\bigwedge\,\frac{a+b}{d}\ngtr 0$, **then**

$$\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx \to \frac{\sqrt{-d\,\text{Sin}[e+f\,x]}}{\sqrt{d\,\text{Sin}[e+f\,x]}}\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{-d\,\text{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[1/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  Sqrt[-d*Sin[e+f*x]]/Sqrt[d*Sin[e+f*x]]*Int[1/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[-d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && NegQ[(a+b)/d]
```

2. $\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

1: $\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\bigwedge\,a^2-b^2\neq 0\,\bigwedge\,c^2-d^2\neq 0\,\bigwedge\,\frac{c+d}{a+b}>0$

- **Note: Alternative antiderivative contributed via email by Martin Welz on 12 April 2014.**

- **Rule: If** $b\,c-a\,d\neq 0\,\bigwedge\,a^2-b^2\neq 0\,\bigwedge\,c^2-d^2\neq 0\,\bigwedge\,\frac{c+d}{a+b}>0$, **then**

$$\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx \to$$

$$\frac{2\,(c+d\,\text{Sin}[e+f\,x])}{f\,(b\,c-a\,d)\,\sqrt{\frac{c+d}{a+b}}\,\text{Cos}[e+f\,x]}\sqrt{\frac{(b\,c-a\,d)\,(1-\text{Sin}[e+f\,x])}{(a+b)\,(c+d\,\text{Sin}[e+f\,x])}}$$

$$\sqrt{-\frac{(b\,c-a\,d)\,(1+\text{Sin}[e+f\,x])}{(a-b)\,(c+d\,\text{Sin}[e+f\,x])}}\;\text{EllipticF}\Big[\text{ArcSin}\Big[\sqrt{\frac{c+d}{a+b}}\,\frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{c+d\,\text{Sin}[e+f\,x]}}\Big],\,\frac{(a+b)\,(c-d)}{(a-b)\,(c+d)}\Big]$$

$$\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\ \sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx \longrightarrow$$

$$\frac{2\,(1-\mathrm{Sin}[e+f\,x])}{f\sqrt{-\frac{a+b}{a-b}}\ \sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\ \sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\sqrt{\frac{a+b\,\mathrm{Sin}[e+f\,x]}{(a-b)\,(1-\mathrm{Sin}[e+f\,x])}}$$

$$\sqrt{\frac{c+d\,\mathrm{Sin}[e+f\,x]}{(c-d)\,(1-\mathrm{Sin}[e+f\,x])}}\ \mathrm{EllipticF}\Big[\mathrm{ArcSin}\Big[\sqrt{-\frac{a+b}{a-b}}\ \frac{1+\mathrm{Sin}[e+f\,x]}{\mathrm{Cos}[e+f\,x]}\Big],\ \frac{(a-b)\,(c+d)}{(a+b)\,(c-d)}\Big]$$

Program code:

```
Int[1/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  2*(c+d*Sin[e+f*x])/(f*(b*c-a*d)*Rt[(c+d)/(a+b),2]*Cos[e+f*x])*
    Sqrt[(b*c-a*d)*(1-Sin[e+f*x])/((a+b)*(c+d*Sin[e+f*x]))]*
    Sqrt[-(b*c-a*d)*(1+Sin[e+f*x])/((a-b)*(c+d*Sin[e+f*x]))]*
    EllipticF[ArcSin[Rt[(c+d)/(a+b),2]*(Sqrt[a+b*Sin[e+f*x]]/Sqrt[c+d*Sin[e+f*x]])],(a+b)*(c-d)/((a-b)*(c+d))] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && PosQ[(c+d)/(a+b)]
```

2: $\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\ \sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\ \bigwedge\ a^2-b^2\neq 0\ \bigwedge\ c^2-d^2\neq 0\ \bigwedge\ \frac{c+d}{a+b}\ngtr 0$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\sqrt{-F[x]}}{\sqrt{F[x]}} == 0$
- **Rule: If** $b\,c-a\,d\neq 0\ \bigwedge\ a^2-b^2\neq 0\ \bigwedge\ c^2-d^2\neq 0\ \bigwedge\ \frac{c+d}{a+b}\ngtr 0$**, then**

$$\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\ \sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx \longrightarrow \frac{\sqrt{-a-b\,\mathrm{Sin}[e+f\,x]}}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\int \frac{1}{\sqrt{-a-b\,\mathrm{Sin}[e+f\,x]}\ \sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[1/(Sqrt[a_.+b_.*sin[e_.+f_.*x_]]*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  Sqrt[-a-b*Sin[e+f*x]]/Sqrt[a+b*Sin[e+f*x]]*Int[1/(Sqrt[-a-b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && NegQ[(c+d)/(a+b)]
```

9: $\int \frac{(d \operatorname{Sin}[e+f x])^{3/2}}{\sqrt{a+b \operatorname{Sin}[e+f x]}} \, dx$ when $a^2-b^2 \neq 0$

Derivation: Algebraic expansion

- **Basis:** $(d\,z)^{3/2} == -\frac{a\,d\,\sqrt{d\,z}}{2\,b} + \frac{d\,\sqrt{d\,z}\,(a+2\,b\,z)}{2\,b}$

- **Rule: If** $a^2-b^2 \neq 0$, **then**

$$\int \frac{(d \operatorname{Sin}[e+f x])^{3/2}}{\sqrt{a+b \operatorname{Sin}[e+f x]}} \, dx \longrightarrow -\frac{a\,d}{2\,b} \int \frac{\sqrt{d \operatorname{Sin}[e+f x]}}{\sqrt{a+b \operatorname{Sin}[e+f x]}} \, dx + \frac{d}{2\,b} \int \frac{\sqrt{d \operatorname{Sin}[e+f x]}\,(a+2\,b \operatorname{Sin}[e+f x])}{\sqrt{a+b \operatorname{Sin}[e+f x]}} \, dx$$

- **Program code:**

```
Int[(d_.*sin[e_.+f_.*x_])^(3/2)/Sqrt[a_.+b_.*sin[e_.+f_.*x_]],x_Symbol] :=
  -a*d/(2*b)*Int[Sqrt[d*Sin[e+f*x]]/Sqrt[a+b*Sin[e+f*x]],x] +
  d/(2*b)*Int[Sqrt[d*Sin[e+f*x]]*(a+2*b*Sin[e+f*x])/Sqrt[a+b*Sin[e+f*x]],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0]
```

10: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\;\wedge\;a^2-b^2\neq 0\;\wedge\;c^2-d^2\neq 0\;\wedge\;0<m<2\;\wedge\;-1<n<2$

Derivation: Nondegenerate sine recurrence 1b with $A\to a\,c,\;B\to b\,c+a\,d,\;C\to b\,d,\;m\to m-1,\;n\to n-1,\;p\to 0$

Rule: If $b\,c-a\,d\neq 0\;\wedge\;a^2-b^2\neq 0\;\wedge\;c^2-d^2\neq 0\;\wedge\;0<m<2\;\wedge\;-1<n<2$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\;\longrightarrow$$

$$-\frac{b\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m-1}\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{f\,(m+n)}+$$

$$\frac{1}{d\,(m+n)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m-2}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n-1}\,\cdot$$

$$\left(a^2\,c\,d\,(m+n)+b\,d\,(b\,c\,(m-1)+a\,d\,n)+(a\,d\,(2\,b\,c+a\,d)\,(m+n)-b\,d\,(a\,c-b\,d\,(m+n-1)))\,\mathrm{Sin}[e+f\,x]+b\,d\,(b\,c\,n+a\,d\,(2\,m+n-1))\,\mathrm{Sin}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -b*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^n/(f*(m+n)) +
  1/(d*(m+n))*Int[(a+b*Sin[e+f*x])^(m-2)*(c+d*Sin[e+f*x])^(n-1)*
    Simp[a^2*c*d*(m+n)+b*d*(b*c*(m-1)+a*d*n)+
      (a*d*(2*b*c+a*d)*(m+n)-b*d*(a*c-b*d*(m+n-1)))*Sin[e+f*x]+
      b*d*(b*c*n+a*d*(2*m+n-1))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[0,m,2] && LtQ[-1,n,2] && NeQ[m+n,0] &&
  (IntegerQ[m] || IntegersQ[2*m,2*n])
```

11: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,m\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $a+b\,z=\frac{b\,(c+d\,z)}{d}-\frac{b\,c-a\,d}{d}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,m\in\mathbb{Z}^+$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\,\longrightarrow$$
$$\frac{b}{d}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m-1}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+1}\,dx-\frac{b\,c-a\,d}{d}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m-1}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  b/d*Int[(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^(n+1),x] -
  (b*c-a*d)/d*Int[(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,n},x] && NeQ[b*c-a*d,0] && IGtQ[m,0]
```

12. $\int (d\,\mathrm{Sin}[e+f\,x])^n\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2=0\,\wedge\,m\in\mathbb{Z}^-$

1: $\int \frac{(d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+b\,z}=\frac{a}{a^2-b^2\,z^2}-\frac{b\,z}{a^2-b^2\,z^2}$

Rule: If $a^2-b^2\neq 0$**, then**

$$\int \frac{(d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx\,\longrightarrow\,a\int \frac{(d\,\mathrm{Sin}[e+f\,x])^n}{a^2-b^2\,\mathrm{Sin}[e+f\,x]^2}\,dx-\frac{b}{d}\int \frac{(d\,\mathrm{Sin}[e+f\,x])^{n+1}}{a^2-b^2\,\mathrm{Sin}[e+f\,x]^2}\,dx$$

Program code:

```
Int[(d_.*sin[e_.+f_.*x_])^n_./(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  a*Int[(d*Sin[e+f*x])^n/(a^2-b^2*Sin[e+f*x]^2),x] -
  b/d*Int[(d*Sin[e+f*x])^(n+1)/(a^2-b^2*Sin[e+f*x]^2),x] /;
FreeQ[{a,b,d,e,f,n},x] && NeQ[a^2-b^2,0]
```

2: $\int (d\,\text{Sin}[e+f\,x])^n\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0\,\wedge\,m+1\in\mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+b\,z}==\frac{a-b\,z}{a^2-b^2\,z^2}$

Rule: If $a^2-b^2\neq 0\,\wedge\,m+1\in\mathbb{Z}^-$**, then**

$$\int (d\,\text{Sin}[e+f\,x])^n\,(a+b\,\text{Sin}[e+f\,x])^m\,dx\;\longrightarrow\;\int \text{ExpandTrig}\Big[\frac{(d\,\text{Sin}[e+f\,x])^n\,(a-b\,\text{Sin}[e+f\,x])^{-m}}{\left(a^2-b^2\,\text{Sin}[e+f\,x]^2\right)^{-m}},\,x\Big]\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[ExpandTrig[(d*sin[e+f*x])^n*(a-b*sin[e+f*x])^(-m)/(a^2-b^2*sin[e+f*x]^2)^(-m),x],x] /;
FreeQ[{a,b,d,e,f,n},x] && NeQ[a^2-b^2,0] && ILtQ[m,-1]
```

X: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$**, then**

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx\;\longrightarrow\;\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  Unintegrable[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

Rules for integrands of the form $(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c\,(d\,\mathrm{Sin}[e+f\,x])^p)^n$

x: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(d\,\mathrm{Csc}[e+f\,x])^n\,dx$ when $n\notin\mathbb{Z}\wedge m\in\mathbb{Z}$

Derivation: Algebraic normalization

- **Basis: If $m\in\mathbb{Z}$, then $(a+b\,\mathrm{Sin}[z])^m == \frac{d^m\,(b+a\,\mathrm{Csc}[z])^m}{(d\,\mathrm{Csc}[z])^m}$**

Note: Although this rule does not introduce a piecewise constant factor, it is better to stay in the sine/cosine world than the secant/cosecant world.

Rule: If $n\notin\mathbb{Z}\wedge m\in\mathbb{Z}$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(d\,\mathrm{Csc}[e+f\,x])^n\,dx \longrightarrow d^m\int (d\,\mathrm{Csc}[e+f\,x])^{n-m}\,(b+a\,\mathrm{Csc}[e+f\,x])^m\,dx$$

Program code:

```
(* Int[(a_.+b_.*sin[e_.+f_.*x_])^m_.*(d_./sin[e_.+f_.*x_])^n_,x_Symbol] :=
  d^m*Int[(d*Csc[e+f*x])^(n-m)*(b+a*Csc[e+f*x])^m,x] /;
FreeQ[{a,b,d,e,f,n},x] && Not[IntegerQ[n]] && IntegerQ[m] *)
```

```
(* Int[(a_.+b_.*cos[e_.+f_.*x_])^m_.*(d_./cos[e_.+f_.*x_])^n_,x_Symbol] :=
  d^m*Int[(d*Sec[e+f*x])^(n-m)*(b+a*Sec[e+f*x])^m,x] /;
FreeQ[{a,b,d,e,f,n},x] && Not[IntegerQ[n]] && IntegerQ[m] *)
```

1: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c\,(d\,\text{Sin}[e+f\,x])^p)^n\,dx$ when $n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(c\,(d\,\text{Sin}[e+f\,x])^p)^n}{(d\,\text{Sin}[e+f\,x])^{n\,p}} == 0$

Rule: If $n\notin\mathbb{Z}$**, then**

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c\,(d\,\text{Sin}[e+f\,x])^p)^n\,dx \longrightarrow \frac{c^{\text{IntPart}[n]}\,(c\,(d\,\text{Sin}[e+f\,x])^p)^{\text{FracPart}[n]}}{(d\,\text{Sin}[e+f\,x])^{p\,\text{FracPart}[n]}}\int (a+b\,\text{Sin}[e+f\,x])^m\,(d\,\text{Sin}[e+f\,x])^{n\,p}\,dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_.*(c_.*(d_.*sin[e_.+f_.*x_])^p_)^n_,x_Symbol] :=
  c^IntPart[n]*(c*(d*Sin[e + f*x])^p)^FracPart[n]/(d*Sin[e + f*x])^(p*FracPart[n])*
    Int[(a+b*Sin[e+f*x])^m*(d*Sin[e+f*x])^(n*p),x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[n]]
```

```
Int[(a_.+b_.*cos[e_.+f_.*x_])^m_.*(c_.*(d_.*cos[e_.+f_.*x_])^p_)^n_,x_Symbol] :=
  c^IntPart[n]*(c*(d*Cos[e + f*x])^p)^FracPart[n]/(d*Cos[e + f*x])^(p*FracPart[n])*
    Int[(a+b*Cos[e+f*x])^m*(d*Cos[e+f*x])^(n*p),x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[n]]
```

Rules for integrands of the form $(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Csc}[e+f\,x])^n$

1: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Csc}[e+f\,x])^n\,dx$ when $n\in\mathbb{Z}$

Derivation: Algebraic normalization

- **Basis:** $c+d\,\mathrm{Csc}[z]==\frac{d+c\,\mathrm{Sin}[z]}{\mathrm{Sin}[z]}$

- **Rule: If** $n\in\mathbb{Z}$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Csc}[e+f\,x])^n\,dx\;\longrightarrow\;\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,(d+c\,\mathrm{Sin}[e+f\,x])^n}{\mathrm{Sin}[e+f\,x]^n}\,dx$$

- **Program code:**

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[(a+b*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n/Sin[e+f*x]^n,x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IntegerQ[n]
```

2. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Csc}[e+f\,x])^n\,dx$ when $n\notin\mathbb{Z}$

1: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Csc}[e+f\,x])^n\,dx$ when $n\notin\mathbb{Z}\,\land\,m\in\mathbb{Z}$

Derivation: Algebraic normalization

- **Basis:** $a+b\,\mathrm{Sin}[z]==\frac{b+a\,\mathrm{Csc}[z]}{\mathrm{Csc}[z]}$

- **Rule: If** $n\notin\mathbb{Z}\,\land\,m\in\mathbb{Z}$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Csc}[e+f\,x])^n\,dx\;\longrightarrow\;\int \frac{(b+a\,\mathrm{Csc}[e+f\,x])^m\,(c+d\,\mathrm{Csc}[e+f\,x])^n}{\mathrm{Csc}[e+f\,x]^m}\,dx$$

- **Program code:**

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  Int[(b+a*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^n/Csc[e+f*x]^m,x] /;
FreeQ[{a,b,c,d,e,f,n},x] && Not[IntegerQ[n]] && IntegerQ[m]
```

```
Int[(a_+b_.*cos[e_.+f_.*x_])^m_.*(c_+d_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  Int[(b+a*Sec[e+f*x])^m*(c+d*Sec[e+f*x])^n/Sec[e+f*x]^m,x] /;
FreeQ[{a,b,c,d,e,f,n},x] && Not[IntegerQ[n]] && IntegerQ[m]
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Csc}[e+f\,x])^n\,dx$ when $n\notin\mathbb{Z}\wedge m\notin\mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x\frac{(c+d\,\mathrm{Csc}[e+f\,x])^n\,\mathrm{Sin}[e+f\,x]^n}{(d+c\,\mathrm{Sin}[e+f\,x])^n}==0$
- **Rule: If** $n\notin\mathbb{Z}\wedge m\notin\mathbb{Z}$**, then**

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Csc}[e+f\,x])^n\,dx\longrightarrow\frac{(c+d\,\mathrm{Csc}[e+f\,x])^n\,\mathrm{Sin}[e+f\,x]^n}{(d+c\,\mathrm{Sin}[e+f\,x])^n}\int\frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,(d+c\,\mathrm{Sin}[e+f\,x])^n}{\mathrm{Sin}[e+f\,x]^n}\,dx$$

- **Program code:**

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  Sin[e+f*x]^n*(c+d*Csc[e+f*x])^n/(d+c*Sin[e+f*x])^n*Int[(a+b*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n/Sin[e+f*x]^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && Not[IntegerQ[n]] && Not[IntegerQ[m]]
```

```
Int[(a_+b_.*cos[e_.+f_.*x_])^m_*(c_+d_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  Cos[e+f*x]^n*(c+d*Sec[e+f*x])^n/(d+c*Cos[e+f*x])^n*Int[(a+b*Cos[e+f*x])^m*(d+c*Cos[e+f*x])^n/Cos[e+f*x]^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && Not[IntegerQ[n]] && Not[IntegerQ[m]]
```