Rules for integrands of the form $(a + b x^n)^p \operatorname{Sin}[c + d x]$

1: $\int (a + b x^n)^p \operatorname{Sin}[c + d x] \, dx$ when $p \in \mathbb{Z}^+$

- **Derivation: Algebraic expansion**
- **Rule: If $p \in \mathbb{Z}^+$, then**

$$\int (a + b x^n)^p \operatorname{Sin}[c + d x] \, dx \longrightarrow \int \operatorname{Sin}[c + d x] \operatorname{ExpandIntegrand}[(a + b x^n)^p, x] \, dx$$

```
Int[(a_+b_.*x_^n_)^p_.*Sin[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Sin[c+d*x],(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[p,0]
```

```
Int[(a_+b_.*x_^n_)^p_.*Cos[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Cos[c+d*x],(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[p,0]
```

2. $\int (a+b\,x^n)^p\,\mathrm{Sin}[c+d\,x]\,dx$ when $p\in\mathbb{Z}^-\wedge n\in\mathbb{Z}$

1. $\int (a+b\,x^n)^p\,\mathrm{Sin}[c+d\,x]\,dx$ when $p\in\mathbb{Z}^-\wedge n\in\mathbb{Z}^+$

1: $\int (a+b\,x^n)^p\,\mathrm{Sin}[c+d\,x]\,dx$ when $p+1\in\mathbb{Z}^-\wedge n-2\in\mathbb{Z}^+$

- **Derivation: Integration by parts**
- **Basis:** $\partial_x\frac{(a+b\,x^n)^{p+1}}{b\,n\,(p+1)} = x^{n-1}\,(a+b\,x^n)^p$
- **Basis:** $\partial_x\left(x^{-n+1}\,\mathrm{Sin}[c+d\,x]\right) = -(n-1)\,x^{-n}\,\mathrm{Sin}[c+d\,x]+d\,x^{-n+1}\,\mathrm{Cos}[c+d\,x]$
- **Rule: If** $p+1\in\mathbb{Z}^-\wedge n-2\in\mathbb{Z}^+$**, then**

$$\int (a+b\,x^n)^p\,\mathrm{Sin}[c+d\,x]\,dx\,\longrightarrow$$

$$\frac{x^{-n+1}\,(a+b\,x^n)^{p+1}\,\mathrm{Sin}[c+d\,x]}{b\,n\,(p+1)}-\frac{-n+1}{b\,n\,(p+1)}\int x^{-n}\,(a+b\,x^n)^{p+1}\,\mathrm{Sin}[c+d\,x]\,dx-\frac{d}{b\,n\,(p+1)}\int x^{-n+1}\,(a+b\,x^n)^{p+1}\,\mathrm{Cos}[c+d\,x]\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_^n_)^p_*Sin[c_.+d_.*x_],x_Symbol] :=
  x^(-n+1)*(a+b*x^n)^(p+1)*Sin[c+d*x]/(b*n*(p+1)) -
  (-n+1)/(b*n*(p+1))*Int[x^(-n)*(a+b*x^n)^(p+1)*Sin[c+d*x],x] -
  d/(b*n*(p+1))*Int[x^(-n+1)*(a+b*x^n)^(p+1)*Cos[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,-1] && IGtQ[n,2]
```

```
Int[(a_+b_.*x_^n_)^p_*Cos[c_.+d_.*x_],x_Symbol] :=
  x^(-n+1)*(a+b*x^n)^(p+1)*Cos[c+d*x]/(b*n*(p+1)) -
  (-n+1)/(b*n*(p+1))*Int[x^(-n)*(a+b*x^n)^(p+1)*Cos[c+d*x],x] +
  d/(b*n*(p+1))*Int[x^(-n+1)*(a+b*x^n)^(p+1)*Sin[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,-1] && IGtQ[n,2]
```

2: $\int (a+b x^n)^p \operatorname{Sin}[c+d x] \, dx$ when $p \in \mathbb{Z}^- \land n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^- \land n \in \mathbb{Z}^+$, then

$$\int (a+b x^n)^p \operatorname{Sin}[c+d x] \, dx \longrightarrow \int \operatorname{Sin}[c+d x] \operatorname{ExpandIntegrand}[(a+b x^n)^p, x] \, dx$$

```
Int[(a_+b_.*x_^n_)^p_*Sin[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Sin[c+d*x],(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && IGtQ[n,0] && (EqQ[n,2] || EqQ[p,-1])
```

```
Int[(a_+b_.*x_^n_)^p_*Cos[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Cos[c+d*x],(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && IGtQ[n,0] && (EqQ[n,2] || EqQ[p,-1])
```

2: $\int (a+b x^n)^p \operatorname{Sin}[c+d x] \, dx$ when $p \in \mathbb{Z}^- \land n \in \mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: If $p \in \mathbb{Z}$, then $(a+b x^n)^p == x^{n p} (b+a x^{-n})^p$

Rule: If $p \in \mathbb{Z}^- \land n \in \mathbb{Z}^-$, then

$$\int (a+b x^n)^p \operatorname{Sin}[c+d x] \, dx \longrightarrow \int x^{n p} (b+a x^{-n})^p \operatorname{Sin}[c+d x] \, dx$$

```
Int[(a_+b_.*x_^n_)^p_*Sin[c_.+d_.*x_],x_Symbol] :=
  Int[x^(n*p)*(b+a*x^(-n))^p*Sin[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && ILtQ[n,0]
```

```
Int[(a_+b_.*x_^n_)^p_*Cos[c_.+d_.*x_],x_Symbol] :=
  Int[x^(n*p)*(b+a*x^(-n))^p*Cos[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && ILtQ[n,0]
```

X: $\int (a+b\,x^n)^p \,\text{Sin}[c+d\,x]\,dx$

Rule:

$$\int (a+b\,x^n)^p \,\text{Sin}[c+d\,x]\,dx \longrightarrow \int (a+b\,x^n)^p \,\text{Sin}[c+d\,x]\,dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*Sin[c_.+d_.*x_],x_Symbol] :=
  Unintegrable[(a+b*x^n)^p*Sin[c+d*x],x] /;
FreeQ[{a,b,c,d,n,p},x]
```

```
Int[(a_+b_.*x_^n_)^p_*Cos[c_.+d_.*x_],x_Symbol] :=
  Unintegrable[(a+b*x^n)^p*Cos[c+d*x],x] /;
FreeQ[{a,b,c,d,n,p},x]
```

Rules for integrands of the form $(e\,x)^m\,(a+b\,x^n)^p\,\text{Sin}[c+d\,x]$

1: $\int (e\,x)^m\,(a+b\,x^n)^p \,\text{Sin}[c+d\,x]\,dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p \,\text{Sin}[c+d\,x]\,dx \longrightarrow \int \text{Sin}[c+d\,x]\,\text{ExpandIntegrand}\left[(e\,x)^m\,(a+b\,x^n)^p,\,x\right]dx$$

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*Sin[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Sin[c+d*x],(e*x)^m*(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IGtQ[p,0]
```

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*Cos[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Cos[c+d*x],(e*x)^m*(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IGtQ[p,0]
```

2: $\int (e\,x)^m\,(a+b\,x^n)^p\,\text{Sin}[c+d\,x]\,dx$ when $p+1\in\mathbb{Z}^- \wedge m==n-1 \wedge (n\in\mathbb{Z} \vee e>0)$

Derivation: Integration by parts

- **Basis: If** $m==n-1 \wedge (n\in\mathbb{Z} \vee e>0)$**, then** $\partial_x \frac{e^m\,(a+b\,x^n)^{p+1}}{b\,n\,(p+1)} == (e\,x)^m\,(a+b\,x^n)^p$

Rule: If $p+1\in\mathbb{Z}^- \wedge m==n-1 \wedge (n\in\mathbb{Z} \vee e>0)$**, then**

$$\int (e\,x)^m\,(a+b\,x^n)^p\,\text{Sin}[c+d\,x]\,dx \longrightarrow \frac{e^m\,(a+b\,x^n)^{p+1}\,\text{Sin}[c+d\,x]}{b\,n\,(p+1)} - \frac{d\,e^m}{b\,n\,(p+1)}\int (a+b\,x^n)^{p+1}\,\text{Cos}[c+d\,x]\,dx$$

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*Sin[c_.+d_.*x_],x_Symbol] :=
  e^m*(a+b*x^n)^(p+1)*Sin[c+d*x]/(b*n*(p+1)) -
  d*e^m/(b*n*(p+1))*Int[(a+b*x^n)^(p+1)*Cos[c+d*x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && ILtQ[p,-1] && EqQ[m,n-1] && (IntegerQ[n] || GtQ[e,0])
```

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*Cos[c_.+d_.*x_],x_Symbol] :=
  e^m*(a+b*x^n)^(p+1)*Cos[c+d*x]/(b*n*(p+1)) +
  d*e^m/(b*n*(p+1))*Int[(a+b*x^n)^(p+1)*Sin[c+d*x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && ILtQ[p,-1] && EqQ[m,n-1] && (IntegerQ[n] || GtQ[e,0])
```

3. $\int x^m (a+b x^n)^p \operatorname{Sin}[c+d x] \, dx$ when $p \in \mathbb{Z}^- \wedge (m \mid n) \in \mathbb{Z}$

1. $\int x^m (a+b x^n)^p \operatorname{Sin}[c+d x] \, dx$ when $p \in \mathbb{Z}^- \wedge n \in \mathbb{Z}^+$

1: $\int x^m (a+b x^n)^p \operatorname{Sin}[c+d x] \, dx$ when $p+1 \in \mathbb{Z}^- \wedge n \in \mathbb{Z}^+ \wedge (m-n+1>0 \vee n>2)$

- **Derivation: Integration by parts**
- **Basis:** $\partial_x \frac{(a+b x^n)^{p+1}}{b n (p+1)} = x^{n-1} (a+b x^n)^p$
- **Basis:** $\partial_x \left(x^{m-n+1} \operatorname{Sin}[c+d x]\right) = (m-n+1) x^{m-n} \operatorname{Sin}[c+d x] + d x^{m-n+1} \operatorname{Cos}[c+d x]$
- **Rule: If** $p+1 \in \mathbb{Z}^- \wedge n \in \mathbb{Z}^+ \wedge (m-n+1>0 \vee n>2)$**, then**

$$\int x^m (a+b x^n)^p \operatorname{Sin}[c+d x] \, dx \longrightarrow$$

$$\frac{x^{m-n+1} (a+b x^n)^{p+1} \operatorname{Sin}[c+d x]}{b n (p+1)} - \frac{m-n+1}{b n (p+1)} \int x^{m-n} (a+b x^n)^{p+1} \operatorname{Sin}[c+d x] \, dx - \frac{d}{b n (p+1)} \int x^{m-n+1} (a+b x^n)^{p+1} \operatorname{Cos}[c+d x] \, dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*Sin[c_.+d_.*x_],x_Symbol] :=
  x^(m-n+1)*(a+b*x^n)^(p+1)*Sin[c+d*x]/(b*n*(p+1)) -
  (m-n+1)/(b*n*(p+1))*Int[x^(m-n)*(a+b*x^n)^(p+1)*Sin[c+d*x],x] -
  d/(b*n*(p+1))*Int[x^(m-n+1)*(a+b*x^n)^(p+1)*Cos[c+d*x],x] /;
FreeQ[{a,b,c,d,m},x] && ILtQ[p,-1] && IGtQ[n,0] && (GtQ[m-n+1,0] || GtQ[n,2]) && RationalQ[m]
```

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*Cos[c_.+d_.*x_],x_Symbol] :=
  x^(m-n+1)*(a+b*x^n)^(p+1)*Cos[c+d*x]/(b*n*(p+1)) -
  (m-n+1)/(b*n*(p+1))*Int[x^(m-n)*(a+b*x^n)^(p+1)*Cos[c+d*x],x] +
  d/(b*n*(p+1))*Int[x^(m-n+1)*(a+b*x^n)^(p+1)*Sin[c+d*x],x] /;
FreeQ[{a,b,c,d,m},x] && ILtQ[p,-1] && IGtQ[n,0] && (GtQ[m-n+1,0] || GtQ[n,2]) && RationalQ[m]
```

2: $\int x^m (a+b x^n)^p \operatorname{Sin}[c+d x] \, dx$ when $p \in \mathbb{Z}^- \wedge n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^- \wedge n \in \mathbb{Z}^+$, then

$$\int x^m (a+b x^n)^p \operatorname{Sin}[c+d x] \, dx \longrightarrow \int \operatorname{Sin}[c+d x] \operatorname{ExpandIntegrand}[x^m (a+b x^n)^p, x] \, dx$$

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*Sin[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Sin[c+d*x],x^m*(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d,m},x] && ILtQ[p,0] && IGtQ[n,0] && (EqQ[n,2] || EqQ[p,-1]) && IntegerQ[m]
```

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*Cos[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Cos[c+d*x],x^m*(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d,m},x] && ILtQ[p,0] && IGtQ[n,0] && (EqQ[n,2] || EqQ[p,-1]) && IntegerQ[m]
```

2: $\int x^m (a+b x^n)^p \operatorname{Sin}[c+d x] \, dx$ when $p \in \mathbb{Z}^- \wedge n \in \mathbb{Z}^-$

- **Derivation: Algebraic simplification**
- **Basis: If $p \in \mathbb{Z}$, then $(a+b x^n)^p == x^{n p} (b+a x^{-n})^p$**
- **Rule: If $p \in \mathbb{Z}^- \wedge n \in \mathbb{Z}^-$, then**

$$\int x^m (a+b x^n)^p \operatorname{Sin}[c+d x] \, dx \longrightarrow \int x^{m+n p} (b+a x^{-n})^p \operatorname{Sin}[c+d x] \, dx$$

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*Sin[c_.+d_.*x_],x_Symbol] :=
  Int[x^(m+n*p)*(b+a*x^(-n))^p*Sin[c+d*x],x] /;
FreeQ[{a,b,c,d,m},x] && ILtQ[p,0] && ILtQ[n,0]
```

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*Cos[c_.+d_.*x_],x_Symbol] :=
  Int[x^(m+n*p)*(b+a*x^(-n))^p*Cos[c+d*x],x] /;
FreeQ[{a,b,c,d,m},x] && ILtQ[p,0] && ILtQ[n,0]
```

X: $\int (e\,x)^m\,(a+b\,x^n)^p\,\text{Sin}[c+d\,x]\,dx$

- **Rule:**

$$\int (e\,x)^m\,(a+b\,x^n)^p\,\text{Sin}[c+d\,x]\,dx \longrightarrow \int (e\,x)^m\,(a+b\,x^n)^p\,\text{Sin}[c+d\,x]\,dx$$

- **Program code:**

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*Sin[c_.+d_.*x_],x_Symbol] :=
  Unintegrable[(e*x)^m*(a+b*x^n)^p*Sin[c+d*x],x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*Cos[c_.+d_.*x_],x_Symbol] :=
  Unintegrable[(e*x)^m*(a+b*x^n)^p*Cos[c+d*x],x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```