# Rules for integrands of the form $(d + e x)^q (a + b \text{ArcTan}[c x])^p$

1. $\int (d + e x)^q (a + b \text{ArcTan}[c x])^p \, dx$ when $p \in \mathbb{Z}^+$

   1. $\int \frac{(a + b \text{ArcTan}[c x])^p}{d + e x} \, dx$ when $p \in \mathbb{Z}^+$

      1: $\int \frac{(a + b \text{ArcTan}[c x])^p}{d + e x} \, dx$ when $p \in \mathbb{Z}^+ \bigwedge c^2 d^2 + e^2 == 0$

- **Derivation: Integration by parts**
- **Basis:** $\frac{1}{d+e x} == -\frac{1}{e} \partial_x \text{Log}\left[\frac{2}{1+\frac{e x}{d}}\right]$
- **Rule: If** $p \in \mathbb{Z}^+ \bigwedge c^2 d^2 + e^2 == 0$**, then**

$$\int \frac{(a + b \text{ArcTan}[c x])^p}{d + e x} \, dx \longrightarrow -\frac{(a + b \text{ArcTan}[c x])^p \text{Log}\left[\frac{2}{1+\frac{e x}{d}}\right]}{e} + \frac{b c p}{e} \int \frac{(a + b \text{ArcTan}[c x])^{p-1} \text{Log}\left[\frac{2}{1+\frac{e x}{d}}\right]}{1 + c^2 x^2} \, dx$$

- **Program code:**

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_./(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcTan[c*x])^p*Log[2/(1+e*x/d)]/e +
  b*c*p/e*Int[(a+b*ArcTan[c*x])^(p-1)*Log[2/(1+e*x/d)]/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && EqQ[c^2*d^2+e^2,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_./(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcCot[c*x])^p*Log[2/(1+e*x/d)]/e -
  b*c*p/e*Int[(a+b*ArcCot[c*x])^(p-1)*Log[2/(1+e*x/d)]/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && EqQ[c^2*d^2+e^2,0]
```

2. $\int \frac{(a+b \operatorname{ArcTan}[c x])^p}{d+e x} \, dx$ when $p \in \mathbb{Z}^+ \wedge c^2 d^2 + e^2 \neq 0$

1: $\int \frac{a+b \operatorname{ArcTan}[c x]}{d+e x} \, dx$ when $c^2 d^2 + e^2 \neq 0$

- **Derivation: Algebraic expansion and integration by parts**
- **Basis:** $\frac{1}{d+e x} == \frac{c}{e (i+c x)} - \frac{c d - i e}{e (i+c x)(d+e x)}$
- **Basis:** $\frac{1}{i+c x} == -\frac{1}{c} \partial_x \operatorname{Log}\left[\frac{2}{1-i c x}\right]$
- **Basis:** $\frac{1}{(i+c x)(d+e x)} == -\frac{1}{c d - i e} \partial_x \operatorname{Log}\left[\frac{2 c (d+e x)}{(c d + i e)(1-i c x)}\right]$
- **Basis:** $\partial_x (a+b \operatorname{ArcTan}[c x]) == \frac{b c}{1+c^2 x^2}$
- **Rule: If** $c^2 d^2 + e^2 \neq 0$**, then**

$$\int \frac{a+b \operatorname{ArcTan}[c x]}{d+e x} \, dx \longrightarrow \frac{c}{e} \int \frac{a+b \operatorname{ArcTan}[c x]}{i+c x} \, dx - \frac{c d - i e}{e} \int \frac{a+b \operatorname{ArcTan}[c x]}{(i+c x)(d+e x)} \, dx \longrightarrow$$

$$-\frac{(a+b \operatorname{ArcTan}[c x]) \operatorname{Log}\left[\frac{2}{1-i c x}\right]}{e} + \frac{b c}{e} \int \frac{\operatorname{Log}\left[\frac{2}{1-i c x}\right]}{1+c^2 x^2} \, dx + \frac{(a+b \operatorname{ArcTan}[c x]) \operatorname{Log}\left[\frac{2 c (d+e x)}{(c d+i e)(1-i c x)}\right]}{e} - \frac{b c}{e} \int \frac{\operatorname{Log}\left[\frac{2 c (d+e x)}{(c d+i e)(1-i c x)}\right]}{1+c^2 x^2} \, dx \longrightarrow$$

$$-\frac{(a+b \operatorname{ArcTan}[c x]) \operatorname{Log}\left[\frac{2}{1-i c x}\right]}{e} + \frac{i b \operatorname{PolyLog}\left[2, 1-\frac{2}{1-i c x}\right]}{2 e} + \frac{(a+b \operatorname{ArcTan}[c x]) \operatorname{Log}\left[\frac{2 c (d+e x)}{(c d+i e)(1-i c x)}\right]}{e} - \frac{i b \operatorname{PolyLog}\left[2, 1-\frac{2 c (d+e x)}{(c d+i e)(1-i c x)}\right]}{2 e}$$

- **Program code:**

```
Int[(a_.+b_.*ArcTan[c_.*x_])/(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcTan[c*x])*Log[2/(1-I*c*x)]/e +
  b*c/e*Int[Log[2/(1-I*c*x)]/(1+c^2*x^2),x] +
  (a+b*ArcTan[c*x])*Log[2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/e -
  b*c/e*Int[Log[2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d^2+e^2,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])/(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcCot[c*x])*Log[2/(1-I*c*x)]/e -
  b*c/e*Int[Log[2/(1-I*c*x)]/(1+c^2*x^2),x] +
  (a+b*ArcCot[c*x])*Log[2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/e +
  b*c/e*Int[Log[2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d^2+e^2,0]
```

**2:** $\int \frac{(a+b\,\mathrm{ArcTan}[c\,x])^2}{d+e\,x}\,dx$ when $c^2 d^2+e^2\neq 0$

**Derivation: Algebraic expansion and integration by parts**

- **Basis:** $\frac{1}{d+e\,x} = \frac{c}{e\,(i+c\,x)} - \frac{c\,d-i\,e}{e\,(i+c\,x)\,(d+e\,x)}$
- **Basis:** $\frac{1}{i+c\,x} = -\frac{1}{c}\,\partial_x \mathrm{Log}\left[\frac{2}{1-i\,c\,x}\right]$
- **Basis:** $\frac{1}{(i+c\,x)\,(d+e\,x)} = -\frac{1}{c\,d-i\,e}\,\partial_x \mathrm{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+i\,e)\,(1-i\,c\,x)}\right]$
- **Basis:** $\partial_x\,(a+b\,\mathrm{ArcTan}[c\,x])^2 = \frac{2\,b\,c\,(a+b\,\mathrm{ArcTan}[c\,x])}{1+c^2\,x^2}$
- **Rule: If** $c^2 d^2+e^2\neq 0$**, then**

$$\int \frac{(a+b\,\mathrm{ArcTan}[c\,x])^2}{d+e\,x}\,dx \to \frac{c}{e}\int \frac{(a+b\,\mathrm{ArcTan}[c\,x])^2}{i+c\,x}\,dx - \frac{c\,d-i\,e}{e}\int \frac{(a+b\,\mathrm{ArcTan}[c\,x])^2}{(i+c\,x)\,(d+e\,x)}\,dx \to$$

$$-\frac{(a+b\,\mathrm{ArcTan}[c\,x])^2\,\mathrm{Log}\left[\frac{2}{1-i\,c\,x}\right]}{e} + \frac{2\,b\,c}{e}\int \frac{(a+b\,\mathrm{ArcTan}[c\,x])\,\mathrm{Log}\left[\frac{2}{1-i\,c\,x}\right]}{1+c^2\,x^2}\,dx +$$
$$\frac{(a+b\,\mathrm{ArcTan}[c\,x])^2\,\mathrm{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+i\,e)\,(1-i\,c\,x)}\right]}{e} - \frac{2\,b\,c}{e}\int \frac{(a+b\,\mathrm{ArcTan}[c\,x])\,\mathrm{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+i\,e)\,(1-i\,c\,x)}\right]}{1+c^2\,x^2}\,dx \to$$

$$-\frac{(a+b\,\mathrm{ArcTan}[c\,x])^2\,\mathrm{Log}\left[\frac{2}{1-i\,c\,x}\right]}{e} + \frac{i\,b\,(a+b\,\mathrm{ArcTan}[c\,x])\,\mathrm{PolyLog}\left[2,1-\frac{2}{1-i\,c\,x}\right]}{e} - \frac{b^2\,\mathrm{PolyLog}\left[3,1-\frac{2}{1-i\,c\,x}\right]}{2\,e} +$$
$$\frac{(a+b\,\mathrm{ArcTan}[c\,x])^2\,\mathrm{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+i\,e)\,(1-i\,c\,x)}\right]}{e} - \frac{i\,b\,(a+b\,\mathrm{ArcTan}[c\,x])\,\mathrm{PolyLog}\left[2,1-\frac{2\,c\,(d+e\,x)}{(c\,d+i\,e)\,(1-i\,c\,x)}\right]}{e} + \frac{b^2\,\mathrm{PolyLog}\left[3,1-\frac{2\,c\,(d+e\,x)}{(c\,d+i\,e)\,(1-i\,c\,x)}\right]}{2\,e}$$

**Program code:**

```
Int[(a_.+b_.*ArcTan[c_.*x_])^2/(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcTan[c*x])^2*Log[2/(1-I*c*x)]/e +
  I*b*(a+b*ArcTan[c*x])*PolyLog[2,1-2/(1-I*c*x)]/e -
  b^2*PolyLog[3,1-2/(1-I*c*x)]/(2*e) +
   (a+b*ArcTan[c*x])^2*Log[2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/e -
  I*b*(a+b*ArcTan[c*x])*PolyLog[2,1-2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/e +
  b^2*PolyLog[3,1-2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/(2*e) /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d^2+e^2,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^2/(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcCot[c*x])^2*Log[2/(1-I*c*x)]/e -
  I*b*(a+b*ArcCot[c*x])*PolyLog[2,1-2/(1-I*c*x)]/e -
  b^2*PolyLog[3,1-2/(1-I*c*x)]/(2*e) +
   (a+b*ArcCot[c*x])^2*Log[2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/e +
  I*b*(a+b*ArcCot[c*x])*PolyLog[2,1-2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/e +
  b^2*PolyLog[3,1-2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/(2*e) /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d^2+e^2,0]
```

3: $\int \frac{(a+b\,\text{ArcTan}[c\,x])^3}{d+e\,x}\,dx$ when $c^2\,d^2+e^2\neq 0$

**Derivation: Algebraic expansion and integration by parts**

- **Basis:** $\frac{1}{d+e\,x} == \frac{c}{e\,(i+c\,x)} - \frac{c\,d-i\,e}{e\,(i+c\,x)\,(d+e\,x)}$
- **Basis:** $\frac{1}{i+c\,x} == -\frac{1}{c}\,\partial_x \text{Log}\left[\frac{2}{1-i\,c\,x}\right]$
- **Basis:** $\frac{1}{(i+c\,x)\,(d+e\,x)} == -\frac{1}{c\,d-i\,e}\,\partial_x \text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+i\,e)\,(1-i\,c\,x)}\right]$
- **Basis:** $\partial_x\,(a+b\,\text{ArcTan}[c\,x])^3 == \frac{3\,b\,c\,(a+b\,\text{ArcTan}[c\,x])^2}{1+c^2\,x^2}$

**Rule: If** $c^2\,d^2+e^2\neq 0$**, then**

$$\int \frac{(a+b\,\text{ArcTan}[c\,x])^3}{d+e\,x}\,dx \rightarrow \frac{c}{e}\int \frac{(a+b\,\text{ArcTan}[c\,x])^3}{i+c\,x}\,dx - \frac{c\,d-i\,e}{e}\int \frac{(a+b\,\text{ArcTan}[c\,x])^3}{(i+c\,x)\,(d+e\,x)}\,dx \rightarrow$$

$$-\frac{(a+b\,\text{ArcTan}[c\,x])^3\,\text{Log}\left[\frac{2}{1-i\,c\,x}\right]}{e} + \frac{3\,b\,c}{e}\int \frac{(a+b\,\text{ArcTan}[c\,x])^2\,\text{Log}\left[\frac{2}{1-i\,c\,x}\right]}{1+c^2\,x^2}\,dx +$$
$$\frac{(a+b\,\text{ArcTan}[c\,x])^3\,\text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+i\,e)\,(1-i\,c\,x)}\right]}{e} - \frac{3\,b\,c}{e}\int \frac{(a+b\,\text{ArcTan}[c\,x])^2\,\text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+i\,e)\,(1-i\,c\,x)}\right]}{1+c^2\,x^2}\,dx \rightarrow$$

$$-\frac{(a+b\,\text{ArcTan}[c\,x])^3\,\text{Log}\left[\frac{2}{1-i\,c\,x}\right]}{e} + \frac{3\,i\,b\,(a+b\,\text{ArcTan}[c\,x])^2\,\text{PolyLog}\left[2,\,1-\frac{2}{1-i\,c\,x}\right]}{2\,e} -$$
$$\frac{3\,b^2\,(a+b\,\text{ArcTan}[c\,x])\,\text{PolyLog}\left[3,\,1-\frac{2}{1-i\,c\,x}\right]}{2\,e} - \frac{3\,i\,b^3\,\text{PolyLog}\left[4,\,1-\frac{2}{1-i\,c\,x}\right]}{4\,e} +$$
$$\frac{(a+b\,\text{ArcTan}[c\,x])^3\,\text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+i\,e)\,(1-i\,c\,x)}\right]}{e} - \frac{3\,i\,b\,(a+b\,\text{ArcTan}[c\,x])^2\,\text{PolyLog}\left[2,\,1-\frac{2\,c\,(d+e\,x)}{(c\,d+i\,e)\,(1-i\,c\,x)}\right]}{2\,e} +$$

$$\frac{3\,b^2\,(a+b\,\text{ArcTan}[c\,x])\,\text{PolyLog}\left[3,\,1-\frac{2\,c\,(d+e\,x)}{(c\,d+i\,e)\,(1-i\,c\,x)}\right]}{2\,e}+\frac{3\,i\,b^3\,\text{PolyLog}\left[4,\,1-\frac{2\,c\,(d+e\,x)}{(c\,d+i\,e)\,(1-i\,c\,x)}\right]}{4\,e}$$

**Program code:**

```
Int[(a_.+b_.*ArcTan[c_.*x_])^3/(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcTan[c*x])^3*Log[2/(1-I*c*x)]/e +
  3*I*b*(a+b*ArcTan[c*x])^2*PolyLog[2,1-2/(1-I*c*x)]/(2*e) -
  3*b^2*(a+b*ArcTan[c*x])*PolyLog[3,1-2/(1-I*c*x)]/(2*e) -
  3*I*b^3*PolyLog[4,1-2/(1-I*c*x)]/(4*e) +
  (a+b*ArcTan[c*x])^3*Log[2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/e -
  3*I*b*(a+b*ArcTan[c*x])^2*PolyLog[2,1-2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/(2*e) +
  3*b^2*(a+b*ArcTan[c*x])*PolyLog[3,1-2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/(2*e) +
  3*I*b^3*PolyLog[4,1-2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/(4*e) /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d^2+e^2,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^3/(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcCot[c*x])^3*Log[2/(1-I*c*x)]/e -
  3*I*b*(a+b*ArcCot[c*x])^2*PolyLog[2,1-2/(1-I*c*x)]/(2*e) -
  3*b^2*(a+b*ArcCot[c*x])*PolyLog[3,1-2/(1-I*c*x)]/(2*e) +
  3*I*b^3*PolyLog[4,1-2/(1-I*c*x)]/(4*e) +
  (a+b*ArcCot[c*x])^3*Log[2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/e +
  3*I*b*(a+b*ArcCot[c*x])^2*PolyLog[2,1-2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/(2*e) +
  3*b^2*(a+b*ArcCot[c*x])*PolyLog[3,1-2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/(2*e) -
  3*I*b^3*PolyLog[4,1-2*c*(d+e*x)/((c*d+I*e)*(1-I*c*x))]/(4*e) /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d^2+e^2,0]
```

2: $\int (d+e\,x)^q\,(a+b\,\mathrm{ArcTan}[c\,x])\,dx$ when $q\neq -1$

**Derivation: Integration by parts**

**Rule: If $q\neq -1$, then**

$$\int (d+e\,x)^q\,(a+b\,\mathrm{ArcTan}[c\,x])\,dx\;\longrightarrow\;\frac{(d+e\,x)^{q+1}\,(a+b\,\mathrm{ArcTan}[c\,x])}{e\,(q+1)}-\frac{b\,c}{e\,(q+1)}\int\frac{(d+e\,x)^{q+1}}{1+c^2\,x^2}\,dx$$

**Program code:**

```
Int[(d_+e_.*x_)^q_.*(a_.+b_.*ArcTan[c_.*x_]),x_Symbol] :=
  (d+e*x)^(q+1)*(a+b*ArcTan[c*x])/(e*(q+1)) -
  b*c/(e*(q+1))*Int[(d+e*x)^(q+1)/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e,q},x] && NeQ[q,-1]
```

```
Int[(d_+e_.*x_)^q_.*(a_.+b_.*ArcCot[c_.*x_]),x_Symbol] :=
  (d+e*x)^(q+1)*(a+b*ArcCot[c*x])/(e*(q+1)) +
  b*c/(e*(q+1))*Int[(d+e*x)^(q+1)/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e,q},x] && NeQ[q,-1]
```

3: $\int (d+e\,x)^q\,(a+b\,\mathrm{ArcTan}[c\,x])^p\,dx$ when $p-1\in\mathbb{Z}^+\wedge q\in\mathbb{Z}\wedge q\neq -1$

**Derivation: Integration by parts**

**Rule: If $p-1\in\mathbb{Z}^+\wedge q\in\mathbb{Z}\wedge q\neq -1$, then**

$$\int (d+e\,x)^q\,(a+b\,\mathrm{ArcTan}[c\,x])^p\,dx\;\longrightarrow$$

$$\frac{(d+e\,x)^{q+1}\,(a+b\,\mathrm{ArcTan}[c\,x])^p}{e\,(q+1)}-\frac{b\,c\,p}{e\,(q+1)}\int (a+b\,\mathrm{ArcTan}[c\,x])^{p-1}\,\mathrm{ExpandIntegrand}\Big[\frac{(d+e\,x)^{q+1}}{1+c^2\,x^2},\,x\Big]\,dx$$

**Program code:**

```
Int[(d_+e_.*x_)^q_.*(a_.+b_.*ArcTan[c_.*x_])^p_,x_Symbol] :=
  (d+e*x)^(q+1)*(a+b*ArcTan[c*x])^p/(e*(q+1)) -
  b*c*p/(e*(q+1))*Int[ExpandIntegrand[(a+b*ArcTan[c*x])^(p-1),(d+e*x)^(q+1)/(1+c^2*x^2),x],x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,1] && IntegerQ[q] && NeQ[q,-1]
```

```
Int[(d_+e_.*x_)^q_.*(a_.+b_.*ArcCot[c_.*x_])^p_,x_Symbol] :=
  (d+e*x)^(q+1)*(a+b*ArcCot[c*x])^p/(e*(q+1)) +
  b*c*p/(e*(q+1))*Int[ExpandIntegrand[(a+b*ArcCot[c*x])^(p-1),(d+e*x)^(q+1)/(1+c^2*x^2),x],x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,1] && IntegerQ[q] && NeQ[q,-1]
```

2. $\int (d + e x)^m (a + b \text{ArcTan}[c x^n])^p \, dx$ when $p \in \mathbb{Z}^+$

  1. $\int (d + e x)^m (a + b \text{ArcTan}[c x^n]) \, dx$

    1. $\int \frac{a + b \text{ArcTan}[c x^n]}{d + e x} \, dx$

      1: $\int \frac{a + b \text{ArcTan}[c x^n]}{d + e x} \, dx$ when $n \in \mathbb{Z}$

**Derivation: Integration by parts**

**Basis:** $\partial_x (a + b \text{ArcTan}[c x^n]) = b c n \frac{x^{n-1}}{1+c^2 x^{2n}}$

**Rule: If** $n \in \mathbb{Z}$, **then**

$$\int \frac{a + b \text{ArcTan}[c x^n]}{d + e x} \, dx \longrightarrow \frac{\text{Log}[d + e x] (a + b \text{ArcTan}[c x^n])}{e} - \frac{b c n}{e} \int \frac{x^{n-1} \text{Log}[d + e x]}{1 + c^2 x^{2n}} \, dx$$

**Program code:**

```
Int[(a_.+b_.*ArcTan[c_.*x_^n_])/(d_+e_.*x_),x_Symbol] :=
  Log[d+e*x]*(a+b*ArcTan[c*x^n])/e -
  b*c*n/e*Int[x^(n-1)*Log[d+e*x]/(1+c^2*x^(2*n)),x] /;
FreeQ[{a,b,c,d,e,n},x] && IntegerQ[n]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_^n_])/(d_+e_.*x_),x_Symbol] :=
  Log[d+e*x]*(a+b*ArcCot[c*x^n])/e +
  b*c*n/e*Int[x^(n-1)*Log[d+e*x]/(1+c^2*x^(2*n)),x] /;
FreeQ[{a,b,c,d,e,n},x] && IntegerQ[n]
```

2: $\int \frac{a + b \,\text{ArcTan}[c\, x^n]}{d + e\, x} \, dx$ when $n \in \mathbb{F}$

**Derivation: Integration by substitution**

**Basis:** If $k \in \mathbb{Z}^+$, then $F[x] == k\, \text{Subst}\left[x^{k-1} F\left[x^k\right],\ x,\ x^{1/k}\right] \partial_x x^{1/k}$

**Rule:** If $n \in \mathbb{F}$, let $k \to \text{Denominator}[n]$, then

$$\int \frac{a + b \,\text{ArcTan}[c\, x^n]}{d + e\, x} \, dx \longrightarrow k\, \text{Subst}\left[\int \frac{x^{k-1} \left(a + b \,\text{ArcTan}\left[c\, x^{k\,n}\right]\right)}{d + e\, x^k} \, dx,\ x,\ x^{1/k}\right]$$

**Program code:**

```
Int[(a_.+b_.*ArcTan[c_.*x_^n_])/(d_+e_.*x_),x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k-1)*(a+b*ArcTan[c*x^(k*n)])/(d+e*x^k),x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d,e},x] && FractionQ[n]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_^n_])/(d_+e_.*x_),x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k-1)*(a+b*ArcCot[c*x^(k*n)])/(d+e*x^k),x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d,e},x] && FractionQ[n]
```

2: $\int (d + e\, x)^m \left(a + b \,\text{ArcTan}[c\, x^n]\right) dx$ when $m \neq -1$

**Derivation: Integration by parts**

**Basis:** $\partial_x \left(a + b \,\text{ArcTan}[c\, x^n]\right) == b\, c\, n\, \frac{x^{n-1}}{1+c^2\, x^{2\,n}}$

**Rule:** If $m \neq -1$, then

$$\int (d + e\, x)^m \left(a + b \,\text{ArcTan}[c\, x^n]\right) dx \longrightarrow \frac{(d + e\, x)^{m+1} \left(a + b \,\text{ArcTan}[c\, x^n]\right)}{e\, (m + 1)} - \frac{b\, c\, n}{e\, (m + 1)} \int \frac{x^{n-1}\, (d + e\, x)^{m+1}}{1 + c^2\, x^{2\,n}} \, dx$$

**Program code:**

```
Int[(d_+e_.*x_)^m_.*(a_.+b_.*ArcTan[c_.*x_^n_]),x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*ArcTan[c*x^n])/(e*(m+1)) -
  b*c*n/(e*(m+1))*Int[x^(n-1)*(d+e*x)^(m+1)/(1+c^2*x^(2*n)),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[m,-1]
```

```
Int[(d_+e_.*x_)^m_.*(a_.+b_.*ArcCot[c_.*x_^n_]),x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*ArcCot[c*x^n])/(e*(m+1)) +
  b*c*n/(e*(m+1))*Int[x^(n-1)*(d+e*x)^(m+1)/(1+c^2*x^(2*n)),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[m,-1]
```

**2:** $\int (d+e\,x)^m\,(a+b\,\text{ArcTan}[c\,x^n])^p\,dx$ when $p-1\in\mathbb{Z}^+\wedge m\in\mathbb{Z}^+$

- **Derivation: Algebraic expansion**
- **Rule: If $p-1\in\mathbb{Z}^+\wedge m\in\mathbb{Z}^+$, then**

$$\int (d+e\,x)^m\,(a+b\,\text{ArcTan}[c\,x^n])^p\,dx\longrightarrow\int (a+b\,\text{ArcTan}[c\,x^n])^p\,\text{ExpandIntegrand}[(d+e\,x)^m,\,x]\,dx$$

- **Program code:**

```
Int[(d_+e_.*x_)^m_.*(a_.+b_.*ArcTan[c_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcTan[c*x^n])^p,(d+e*x)^m,x],x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[p,1] && IGtQ[m,0]
```

```
Int[(d_+e_.*x_)^m_.*(a_.+b_.*ArcCot[c_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCot[c*x^n])^p,(d+e*x)^m,x],x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[p,1] && IGtQ[m,0]
```

**U:** $\int (d+e\,x)^m\,(a+b\,\text{ArcTan}[c\,x^n])^p\,dx$

- **Rule:**

$$\int (d+e\,x)^m\,(a+b\,\text{ArcTan}[c\,x^n])^p\,dx\longrightarrow\int (d+e\,x)^m\,(a+b\,\text{ArcTan}[c\,x^n])^p\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcTan[c_.*x_^n_])^p_.,x_Symbol] :=
  Unintegrable[(d+e*x)^m*(a+b*ArcTan[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcCot[c_.*x_^n_])^p_.,x_Symbol] :=
  Unintegrable[(d+e*x)^m*(a+b*ArcCot[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```