

## Rules for integrands involving trig integral functions

1.  $\int \text{SinIntegral}[a + b x] \, dx$

**1:**  $\int \text{SinIntegral}[a + b x] \, dx$

- Derivation: Integration by parts

- Rule:

$$\int \text{SinIntegral}[a + b x] \, dx \rightarrow \frac{(a + b x) \text{SinIntegral}[a + b x]}{b} + \frac{\text{Cos}[a + b x]}{b}$$

- Program code:

```
Int[SinIntegral[a_.+b_.*x_],x_Symbol] :=  
  (a+b*x)*SinIntegral[a+b*x]/b + Cos[a+b*x]/b /;  
FreeQ[{a,b},x]
```

```
Int[CosIntegral[a_.+b_.*x_],x_Symbol] :=  
  (a+b*x)*CosIntegral[a+b*x]/b - Sin[a+b*x]/b /;  
FreeQ[{a,b},x]
```

2.  $\int (c + d x)^m \text{SinIntegral}[a + b x] dx$

1:  $\int \frac{\text{SinIntegral}[b x]}{x} dx$

- Basis:  $\text{SinIntegral}[z] == \frac{1}{2} i (\text{ExpIntegralE}[1, -i z] - \text{ExpIntegralE}[1, i z] + \text{Log}[-i z] - \text{Log}[i z])$
- Basis:  $\text{CosIntegral}[z] == \frac{1}{2} (-\text{ExpIntegralE}[1, -i z] - \text{ExpIntegralE}[1, i z] - \text{Log}[-i z] - \text{Log}[i z] + 2 \text{Log}[z])$
- Rule:

$$\int \frac{\text{SinIntegral}[b x]}{x} dx \rightarrow \frac{1}{2} b x \text{HypergeometricPFQ}[\{1, 1, 1\}, \{2, 2, 2\}, -i b x] + \frac{1}{2} b x \text{HypergeometricPFQ}[\{1, 1, 1\}, \{2, 2, 2\}, i b x]$$

- Program code:

```
Int[SinIntegral[b_.*x_]/x_,x_Symbol] :=
  1/2*b*x*HypergeometricPFQ[{1,1,1},{2,2,2},-I*b*x] +
  1/2*b*x*HypergeometricPFQ[{1,1,1},{2,2,2},I*b*x] /;
FreeQ[b,x]
```

```
Int[CosIntegral[b_.*x_]/x_,x_Symbol] :=
  -1/2*I*b*x*HypergeometricPFQ[{1,1,1},{2,2,2},-I*b*x] +
  1/2*I*b*x*HypergeometricPFQ[{1,1,1},{2,2,2},I*b*x] +
  EulerGamma*Log[x] +
  1/2*Log[b*x]^2 /;
FreeQ[b,x]
```

**2:**  $\int (c + dx)^m \text{SinIntegral}[a + bx] dx$  when  $m \neq -1$

**Derivation: Integration by parts**

**Rule:** If  $m \neq -1$ , then

$$\int (c + dx)^m \text{SinIntegral}[a + bx] dx \rightarrow \frac{(c + dx)^{m+1} \text{SinIntegral}[a + bx]}{d(m+1)} - \frac{b}{d(m+1)} \int \frac{(c + dx)^{m+1} \sin[a + bx]}{a + bx} dx$$

**Program code:**

```
Int[(c_.+d_.*x_)^m_.*SinIntegral[a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*SinIntegral[a+b*x]/(d*(m+1)) -
  b/(d*(m+1))*Int[(c+d*x)^(m+1)*Sin[a+b*x]/(a+b*x),x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1]
```

```
Int[(c_.+d_.*x_)^m_.*CosIntegral[a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*CosIntegral[a+b*x]/(d*(m+1)) -
  b/(d*(m+1))*Int[(c+d*x)^(m+1)*Cos[a+b*x]/(a+b*x),x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1]
```

**2.**  $\int \text{SinIntegral}[a + bx]^2 dx$

**1:**  $\int \text{SinIntegral}[a + bx]^2 dx$

**Derivation: Integration by parts**

**Rule:**

$$\int \text{SinIntegral}[a + bx]^2 dx \rightarrow \frac{(a + bx) \text{SinIntegral}[a + bx]^2}{b} - 2 \int \sin[a + bx] \text{SinIntegral}[a + bx] dx$$

**Program code:**

```
Int[SinIntegral[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*SinIntegral[a+b*x]^2/b -
  2*Int[Sin[a+b*x]*SinIntegral[a+b*x],x] /;
FreeQ[{a,b},x]
```

```
Int[CosIntegral[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*CosIntegral[a+b*x]^2/b -
  2*Int[Cos[a+b*x]*CosIntegral[a+b*x],x] /;
FreeQ[{a,b},x]
```

2.  $\int (c + d x)^m \text{SinIntegral}[a + b x]^2 dx$

**1:**  $\int x^m \text{SinIntegral}[b x]^2 dx$  when  $m \in \mathbb{Z}^+$

**Derivation: Integration by parts**

**Rule: If  $m \in \mathbb{Z}^+$ , then**

$$\int x^m \text{SinIntegral}[b x]^2 dx \rightarrow \frac{x^{m+1} \text{SinIntegral}[b x]^2}{m+1} - \frac{2}{m+1} \int x^m \text{Sin}[b x] \text{SinIntegral}[b x] dx$$

**Program code:**

```
Int[x_^m_.*SinIntegral[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*SinIntegral[b*x]^2/(m+1) -
  2/(m+1)*Int[x^m*Sin[b*x]*SinIntegral[b*x],x] /;
FreeQ[b,x] && IGtQ[m,0]
```

```
Int[x_^m_.*CosIntegral[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*CosIntegral[b*x]^2/(m+1) -
  2/(m+1)*Int[x^m*Cos[b*x]*CosIntegral[b*x],x] /;
FreeQ[b,x] && IGtQ[m,0]
```

**2:**  $\int (c + d x)^m \text{SinIntegral}[a + b x]^2 dx$  when  $m \in \mathbb{Z}^+$

**Derivation: Iterated integration by parts**

**Rule: If  $m \in \mathbb{Z}^+$ , then**

$$\int (c + d x)^m \text{SinIntegral}[a + b x]^2 dx \rightarrow$$

$$\frac{(a + b x) (c + d x)^m \text{SinIntegral}[a + b x]^2}{b (m + 1)} -$$

$$\frac{2}{m+1} \int (c+dx)^m \sin[ax+bx] \operatorname{SinIntegral}[ax+bx] dx + \frac{(bc-ad)m}{b(m+1)} \int (c+dx)^{m-1} \operatorname{SinIntegral}[ax+bx]^2 dx$$

Program code:

```
Int[(c_.+d_.**x_)^m_.*SinIntegral[a_+b_.**x_]^2,x_Symbol] :=
  (a+b*x)*(c+d*x)^m*SinIntegral[a+b*x]^2/(b*(m+1)) -
  2/(m+1)*Int[(c+d*x)^m*Sin[a+b*x]*SinIntegral[a+b*x],x] +
  (b*c-a*d)*m/(b*(m+1))*Int[(c+d*x)^(m-1)*SinIntegral[a+b*x]^2,x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

```
Int[(c_.+d_.**x_)^m_.*CosIntegral[a_+b_.**x_]^2,x_Symbol] :=
  (a+b*x)*(c+d*x)^m*CosIntegral[a+b*x]^2/(b*(m+1)) -
  2/(m+1)*Int[(c+d*x)^m*Cos[a+b*x]*CosIntegral[a+b*x],x] +
  (b*c-a*d)*m/(b*(m+1))*Int[(c+d*x)^(m-1)*CosIntegral[a+b*x]^2,x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

**x:**  $\int x^m \operatorname{SinIntegral}[a+bx]^2 dx$  when  $m+2 \in \mathbb{Z}^-$

Derivation: Inverted integration by parts

Rule: If  $m+2 \in \mathbb{Z}^-$ , then

$$\int x^m \operatorname{SinIntegral}[a+bx]^2 dx \rightarrow \frac{b x^{m+2} \operatorname{SinIntegral}[a+bx]^2}{a(m+1)} + \frac{x^{m+1} \operatorname{SinIntegral}[a+bx]^2}{m+1} - \frac{2b}{a(m+1)} \int x^{m+1} \sin[ax+bx] \operatorname{SinIntegral}[a+bx] dx - \frac{b(m+2)}{a(m+1)} \int x^{m+1} \operatorname{SinIntegral}[a+bx]^2 dx$$

Program code:

```
(* Int[x_^m_.*SinIntegral[a_+b_.**x_]^2,x_Symbol] :=
  b*x^(m+2)*SinIntegral[a+b*x]^2/(a*(m+1)) +
  x^(m+1)*SinIntegral[a+b*x]^2/(m+1) -
  2*b/(a*(m+1))*Int[x^(m+1)*Sin[a+b*x]*SinIntegral[a+b*x],x] -
  b*(m+2)/(a*(m+1))*Int[x^(m+1)*SinIntegral[a+b*x]^2,x] /;
FreeQ[{a,b},x] && ILtQ[m,-2] *)
```

```
(* Int[x_^m_.*CosIntegral[a_+b_.*x_]^2,x_Symbol] :=
  b*x^(m+2)*CosIntegral[a+b*x]^2/(a*(m+1)) +
  x^(m+1)*CosIntegral[a+b*x]^2/(m+1) -
  2*b/(a*(m+1))*Int[x^(m+1)*Cos[a+b*x]*CosIntegral[a+b*x],x] -
  b*(m+2)/(a*(m+1))*Int[x^(m+1)*CosIntegral[a+b*x]^2,x] /;
FreeQ[{a,b},x] && ILtQ[m,-2] *)
```

3.  $\int u \sin[a + bx] \operatorname{SinIntegral}[c + dx] \, dx$

1:  $\int \sin[a + bx] \operatorname{SinIntegral}[c + dx] \, dx$

Reference: G&R 5.32.2

Reference: G&R 5.31.1

Derivation: Integration by parts

Rule:

$$\int \sin[a + bx] \operatorname{SinIntegral}[c + dx] \, dx \rightarrow -\frac{\cos[a + bx] \operatorname{SinIntegral}[c + dx]}{b} + \frac{d}{b} \int \frac{\cos[a + bx] \operatorname{Sin}[c + dx]}{c + dx} \, dx$$

Program code:

```
Int[Sin[a_+b_.*x_]*SinIntegral[c_+d_.*x_],x_Symbol] :=
  -Cos[a+b*x]*SinIntegral[c+d*x]/b +
  d/b*Int[Cos[a+b*x]*Sin[c+d*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d},x]
```

```
Int[Cos[a_+b_.*x_]*CosIntegral[c_+d_.*x_],x_Symbol] :=
  Sin[a+b*x]*CosIntegral[c+d*x]/b -
  d/b*Int[Sin[a+b*x]*Cos[c+d*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d},x]
```

2.  $\int (e + fx)^m \sin[a + bx] \operatorname{SinIntegral}[c + dx] \, dx$

1:  $\int (e + fx)^m \sin[a + bx] \operatorname{SinIntegral}[c + dx] \, dx$  when  $m \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If  $m \in \mathbb{Z}^+$ , then

$$\int (e + f x)^m \sin[a + b x] \operatorname{SinIntegral}[c + d x] dx \rightarrow$$

$$- \frac{(e + f x)^m \cos[a + b x] \operatorname{SinIntegral}[c + d x]}{b} + \frac{d}{b} \int \frac{(e + f x)^m \cos[a + b x] \sin[c + d x]}{c + d x} dx + \frac{f m}{b} \int (e + f x)^{m-1} \cos[a + b x] \operatorname{SinIntegral}[c + d x] dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sin[a_.+b_.*x_] * SinIntegral[c_.+d_.*x_], x_Symbol] :=
  -(e+f*x)^m * Cos[a+b*x] * SinIntegral[c+d*x] / b +
  d/b * Int[(e+f*x)^m * Cos[a+b*x] * Sin[c+d*x] / (c+d*x), x] +
  f*m/b * Int[(e+f*x)^(m-1) * Cos[a+b*x] * SinIntegral[c+d*x], x] /;
FreeQ[{a,b,c,d,e,f}, x] && IGtQ[m, 0]
```

```
Int[(e_.+f_.*x_)^m_.*Cos[a_.+b_.*x_] * CosIntegral[c_.+d_.*x_], x_Symbol] :=
  (e+f*x)^m * Sin[a+b*x] * CosIntegral[c+d*x] / b -
  d/b * Int[(e+f*x)^m * Sin[a+b*x] * Cos[c+d*x] / (c+d*x), x] -
  f*m/b * Int[(e+f*x)^(m-1) * Sin[a+b*x] * CosIntegral[c+d*x], x] /;
FreeQ[{a,b,c,d,e,f}, x] && IGtQ[m, 0]
```

**2:**  $\int (e + f x)^m \sin[a + b x] \operatorname{SinIntegral}[c + d x] dx$  when  $m + 1 \in \mathbb{Z}^-$

Derivation: Inverted integration by parts

Rule: If  $m + 1 \in \mathbb{Z}^-$ , then

$$\int (e + f x)^m \sin[a + b x] \operatorname{SinIntegral}[c + d x] dx \rightarrow$$

$$\frac{(e + f x)^{m+1} \sin[a + b x] \operatorname{SinIntegral}[c + d x]}{f(m+1)} -$$

$$\frac{d}{f(m+1)} \int \frac{(e + f x)^{m+1} \sin[a + b x] \sin[c + d x]}{c + d x} dx - \frac{b}{f(m+1)} \int (e + f x)^{m+1} \cos[a + b x] \operatorname{SinIntegral}[c + d x] dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sin[a_.+b_.*x_] * SinIntegral[c_.+d_.*x_], x_Symbol] :=
  (e+f*x)^(m+1) * Sin[a+b*x] * SinIntegral[c+d*x] / (f*(m+1)) -
  d/(f*(m+1)) * Int[(e+f*x)^(m+1) * Sin[a+b*x] * Sin[c+d*x] / (c+d*x), x] -
  b/(f*(m+1)) * Int[(e+f*x)^(m+1) * Cos[a+b*x] * SinIntegral[c+d*x], x] /;
FreeQ[{a,b,c,d,e,f}, x] && ILtQ[m, -1]
```

```

Int[(e_.+f_.*x_)^m_.*Cos[a_.+b_.*x_]*CosIntegral[c_.+d_.*x_],x_Symbol] :=
  (e+f*x)^(m+1)*Cos[a+b*x]*CosIntegral[c+d*x]/(f*(m+1)) -
  d/(f*(m+1))*Int[(e+f*x)^(m+1)*Cos[a+b*x]*Cos[c+d*x]/(c+d*x),x] +
  b/(f*(m+1))*Int[(e+f*x)^(m+1)*Sin[a+b*x]*CosIntegral[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && ILtQ[m,-1]

```

4.  $\int u \cos[a + bx] \operatorname{SinIntegral}[c + dx] \, dx$

**1:**  $\int \cos[a + bx] \operatorname{SinIntegral}[c + dx] \, dx$

Reference: G&R 5.32.1

Reference: G&R 5.31.2

Derivation: Integration by parts

Rule:

$$\int \cos[a + bx] \operatorname{SinIntegral}[c + dx] \, dx \rightarrow \frac{\sin[a + bx] \operatorname{SinIntegral}[c + dx]}{b} - \frac{d}{b} \int \frac{\sin[a + bx] \sin[c + dx]}{c + dx} \, dx$$

Program code:

```

Int[Cos[a_.+b_.*x_]*SinIntegral[c_.+d_.*x_],x_Symbol] :=
  Sin[a+b*x]*SinIntegral[c+d*x]/b -
  d/b*Int[Sin[a+b*x]*Sin[c+d*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d},x]

```

```

Int[Sin[a_.+b_.*x_]*CosIntegral[c_.+d_.*x_],x_Symbol] :=
  -Cos[a+b*x]*CosIntegral[c+d*x]/b +
  d/b*Int[Cos[a+b*x]*Cos[c+d*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d},x]

```

2.  $\int (e + fx)^m \cos[a + bx] \operatorname{SinIntegral}[c + dx] \, dx$

**1:**  $\int (e + fx)^m \cos[a + bx] \operatorname{SinIntegral}[c + dx] \, dx$  when  $m \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If  $m \in \mathbb{Z}^+$ , then



$$\int (e + f x)^m \cos[a + b x] \operatorname{SinIntegral}[c + d x] dx \rightarrow$$

$$\frac{(e + f x)^m \sin[a + b x] \operatorname{SinIntegral}[c + d x]}{b} - \frac{d}{b} \int \frac{(e + f x)^m \sin[a + b x] \sin[c + d x]}{c + d x} dx - \frac{f m}{b} \int (e + f x)^{m-1} \sin[a + b x] \operatorname{SinIntegral}[c + d x] dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Cos[a_.+b_.*x_]*SinIntegral[c_.+d_.*x_],x_Symbol] :=
  (e+f*x)^m*sin[a+b*x]*SinIntegral[c+d*x]/b -
  d/b*Int[(e+f*x)^m*sin[a+b*x]*Sin[c+d*x]/(c+d*x),x] -
  f*m/b*Int[(e+f*x)^(m-1)*sin[a+b*x]*SinIntegral[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sin[a_.+b_.*x_]*CosIntegral[c_.+d_.*x_],x_Symbol] :=
  -(e+f*x)^m*cos[a+b*x]*CosIntegral[c+d*x]/b +
  d/b*Int[(e+f*x)^m*cos[a+b*x]*Cos[c+d*x]/(c+d*x),x] +
  f*m/b*Int[(e+f*x)^(m-1)*cos[a+b*x]*CosIntegral[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0]
```

**2:**  $\int (e + f x)^m \cos[a + b x] \operatorname{SinIntegral}[c + d x] dx$  when  $m + 1 \in \mathbb{Z}^-$

Derivation: Inverted integration by parts

Rule: If  $m + 1 \in \mathbb{Z}^-$ , then

$$\int (e + f x)^m \cos[a + b x] \operatorname{SinIntegral}[c + d x] dx \rightarrow$$

$$\frac{(e + f x)^{m+1} \cos[a + b x] \operatorname{SinIntegral}[c + d x]}{f(m+1)} -$$

$$\frac{d}{f(m+1)} \int \frac{(e + f x)^{m+1} \cos[a + b x] \sin[c + d x]}{c + d x} dx + \frac{b}{f(m+1)} \int (e + f x)^{m+1} \sin[a + b x] \operatorname{SinIntegral}[c + d x] dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Cos[a_.+b_.*x_]*SinIntegral[c_.+d_.*x_],x_Symbol] :=
  (e+f*x)^(m+1)*Cos[a+b*x]*SinIntegral[c+d*x]/(f*(m+1)) -
  d/(f*(m+1))*Int[(e+f*x)^(m+1)*Cos[a+b*x]*Sin[c+d*x]/(c+d*x),x] +
  b/(f*(m+1))*Int[(e+f*x)^(m+1)*Sin[a+b*x]*SinIntegral[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && ILtQ[m,-1]
```

```

Int[(e_.+f_.*x_)^m_*Sin[a_.+b_.*x_]*CosIntegral[c_.+d_.*x_],x_Symbol] :=
  (e+f*x)^(m+1)*Sin[a+b*x]*CosIntegral[c+d*x]/(f*(m+1)) -
  d/(f*(m+1))*Int[(e+f*x)^(m+1)*Sin[a+b*x]*Cos[c+d*x]/(c+d*x),x] -
  b/(f*(m+1))*Int[(e+f*x)^(m+1)*Cos[a+b*x]*CosIntegral[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && ILtQ[m,-1]

```

5.  $\int \text{SinIntegral}[d(a + b \log[c x^n])] dx$

**1:**  $\int \text{SinIntegral}[d(a + b \log[c x^n])] dx$

**Derivation: Integration by parts**

■ **Basis:**  $\partial_x \text{SinIntegral}[d(a + b \log[c x^n])] = \frac{b d n \text{Sin}[d(a + b \log[c x^n])]}{x(d(a + b \log[c x^n]))}$

**Rule:** If  $m \neq -1$ , then

$$\int \text{SinIntegral}[d(a + b \log[c x^n])] dx \rightarrow x \text{SinIntegral}[d(a + b \log[c x^n])] - b d n \int \frac{\text{Sin}[d(a + b \log[c x^n])]}{d(a + b \log[c x^n])} dx$$

**Program code:**

```

Int[SinIntegral[d_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  x*SinIntegral[d*(a+b*Log[c*x^n])] - b*d*n*Int[Sin[d*(a+b*Log[c*x^n])]/(d*(a+b*Log[c*x^n])),x] /;
FreeQ[{a,b,c,d,n},x]

```

```

Int[CosIntegral[d_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  x*CosIntegral[d*(a+b*Log[c*x^n])] - b*d*n*Int[Cos[d*(a+b*Log[c*x^n])]/(d*(a+b*Log[c*x^n])),x] /;
FreeQ[{a,b,c,d,n},x]

```

**2:**  $\int \frac{\text{SinIntegral}[d(a + b \log[c x^n])]}{x} dx$

Derivation: Integration by substitution

Basis:  $\frac{F[\log[c x^n]]}{x} = \frac{1}{n} \text{Subst}[F[x], x, \log[c x^n]] \partial_x \log[c x^n]$

Rule:

$$\int \frac{\text{SinIntegral}[d(a + b \log[c x^n])]}{x} dx \rightarrow \frac{1}{n} \text{Subst}[\text{SinIntegral}[d(a + b x)], x, \log[c x^n]]$$

Program code:

```
Int[F_[d_.*(a_.+b_.*Log[c_.*x_^n_.])]/x_,x_Symbol] :=
  1/n*Subst[F[d*(a+b*x)],x,Log[c*x^n]] /;
FreeQ[{a,b,c,d,n},x] && MemberQ[{SinIntegral,CosIntegral},x]
```

**3:**  $\int (e x)^m \text{SinIntegral}[d(a + b \log[c x^n])] dx$  when  $m \neq -1$

Derivation: Integration by parts

Basis:  $\partial_x \text{SinIntegral}[d(a + b \log[c x^n])] = \frac{b d n \sin[d(a + b \log[c x^n])]}{x(d(a + b \log[c x^n]))}$

Rule: If  $m \neq -1$ , then

$$\int (e x)^m \text{SinIntegral}[d(a + b \log[c x^n])] dx \rightarrow \frac{(e x)^{m+1} \text{SinIntegral}[d(a + b \log[c x^n])]}{e(m+1)} - \frac{b d n}{m+1} \int \frac{(e x)^m \sin[d(a + b \log[c x^n])]}{d(a + b \log[c x^n])} dx$$

Program code:

```
Int[(e_.*x_)^m_.*SinIntegral[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  (e*x)^(m+1)*SinIntegral[d*(a+b*Log[c*x^n])]/(e*(m+1)) -
  b*d*n/(m+1)*Int[(e*x)^m*SIN[d*(a+b*Log[c*x^n])]/(d*(a+b*Log[c*x^n])),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[m,-1]
```

```
Int[(e_.*x_)^m_.*CosIntegral[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  (e*x)^(m+1)*CosIntegral[d*(a+b*Log[c*x^n])]/(e*(m+1)) -
  b*d*n/(m+1)*Int[(e*x)^m*COS[d*(a+b*Log[c*x^n])]/(d*(a+b*Log[c*x^n])),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[m,-1]
```