Rules for integrands of the form $\left(A+B\operatorname{Log}\left[e\left(\frac{a+b x}{c+d x}\right)^n\right]\right)^p$

1: $\int\left(A+B\operatorname{Log}\left[e\left(\frac{a+b x}{c+d x}\right)^n\right]\right)^p dx$ when $b c-a d \neq 0 \land p \in \mathbb{Z}^+$

- **Derivation: Integration by parts**
- **Basis:** $1 == \partial_x \frac{a+b x}{b}$
- **Basis:** $\partial_x\left(A+B\operatorname{Log}\left[e\left(\frac{a+b x}{c+d x}\right)^n\right]\right)^p == B n p\,(b c-a d)\,\frac{\left(A+B\operatorname{Log}\left[e\left(\frac{a+b x}{c+d x}\right)^n\right]\right)^{p-1}}{(a+b x)\,(c+d x)}$
- **Rule: If** $b c-a d \neq 0 \land p \in \mathbb{Z}^+$**, then**

$$\int\left(A+B\operatorname{Log}\left[e\left(\frac{a+b x}{c+d x}\right)^n\right]\right)^p dx \longrightarrow \frac{(a+b x)\left(A+B\operatorname{Log}\left[e\left(\frac{a+b x}{c+d x}\right)^n\right]\right)^p}{b}-\frac{B n p\,(b c-a d)}{b}\int\frac{\left(A+B\operatorname{Log}\left[e\left(\frac{a+b x}{c+d x}\right)^n\right]\right)^{p-1}}{c+d x}dx$$

- **Program code:**

```
Int[(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])^p_.,x_Symbol] :=
  (a+b*x)*(A+B*Log[e*((a+b*x)/(c+d*x))^n])^p/b -
  B*n*p*(b*c-a*d)/b*Int[(A+B*Log[e*((a+b*x)/(c+d*x))^n])^(p-1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,A,B,n},x] && NeQ[b*c-a*d,0] && IGtQ[p,0]
```

```
Int[(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])^p_.,x_Symbol] :=
  (a+b*x)*(A+B*Log[e*(a+b*x)^n/(c+d*x)^n])^p/b -
  B*n*p*(b*c-a*d)/b*Int[(A+B*Log[e*(a+b*x)^n/(c+d*x)^n])^(p-1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,A,B,n},x] && EqQ[n+mn,0] && NeQ[b*c-a*d,0] && IGtQ[p,0]
```

- **Note: This rule unifies the above two rules, but is inelegant...**

```
(* Int[(A_.+B_.*Log[e_.*((a_.+b_.*x_)^n1_.*(c_.+d_.*x_)^n2_)^n_.])^p_.,x_Symbol] :=
  (a+b*x)*(A+B*Log[e*((a+b*x)^n1/(c+d*x)^n1)^n])^p/b -
  B*n*p*(b*c-a*d)/b*Int[(A+B*Log[e*((a+b*x)^n1/(c+d*x)^n1)^n])^(p-1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,A,B,n},x] && EqQ[n1+n2,0] && GtQ[n1,0] && (EqQ[n1,1] || EqQ[n,1]) && NeQ[b*c-a*d,0] && IGtQ[p,0] *)
```

U: $\int \left(A + B \operatorname{Log}\left[e \left(\frac{a+b x}{c+d x}\right)^n\right]\right)^p dx$

Rule:

$$\int \left(A + B \operatorname{Log}\left[e \left(\frac{a+b x}{c+d x}\right)^n\right]\right)^p dx \longrightarrow \int \left(A + B \operatorname{Log}\left[e \left(\frac{a+b x}{c+d x}\right)^n\right]\right)^p dx$$

Program code:

```
Int[(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])^p_,x_Symbol] :=
  Unintegrable[(A+B*Log[e*((a+b*x)/(c+d*x))^n])^p,x] /;
FreeQ[{a,b,c,d,e,A,B,n,p},x]
```

```
Int[(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])^p_,x_Symbol] :=
  Unintegrable[(A+B*Log[e*(a+b*x)^n/(c+d*x)^n])^p,x] /;
FreeQ[{a,b,c,d,e,A,B,n,p},x] && EqQ[n+mn,0]
```

N: $\int \left(A + B \operatorname{Log}\left[e \left(\frac{u}{v}\right)^n\right]\right)^p dx$ when $u == a + b x \land v == c + d x$

Derivation: Algebraic normalization

Rule: If $u == a + b x \land v == c + d x$**, then**

$$\int \left(A + B \operatorname{Log}\left[e \left(\frac{u}{v}\right)^n\right]\right)^p dx \longrightarrow \int \left(A + B \operatorname{Log}\left[e \left(\frac{a+b x}{c+d x}\right)^n\right]\right)^p dx$$

Program code:

```
Int[(A_.+B_.*Log[e_.*(u_/v_)^n_.])^p_.,x_Symbol] :=
  Int[(A+B*Log[e*(ExpandToSum[u,x]/ExpandToSum[v,x])^n])^p,x] /;
FreeQ[{e,A,B,n,p},x] && LinearQ[{u,v},x] && Not[LinearMatchQ[{u,v},x]]
```

```
Int[(A_.+B_.*Log[e_.*u_^n_.*v_^mn_])^p_.,x_Symbol] :=
  Int[(A+B*Log[e*ExpandToSum[u,x]^n/ExpandToSum[v,x]^n])^p,x] /;
FreeQ[{e,A,B,n,p},x] && EqQ[n+mn,0] && IGtQ[n,0] && LinearQ[{u,v},x] && Not[LinearMatchQ[{u,v},x]]
```

Rules for integrands of the form $(f+g\,x)^m \left(A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p$

1. $\int (f+g\,x)^m \left(A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right) dx$ when $b\,c-a\,d\neq 0$

   1. $\int \frac{A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]}{f+g\,x} dx$ when $b\,c-a\,d\neq 0$

      1: $\int \frac{A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]}{f+g\,x} dx$ when $b\,c-a\,d\neq 0\,\wedge\,b\,f-a\,g=0$

Derivation: Integration by parts

- **Basis:** If $b\,f-a\,g=0$, then $\frac{1}{f+g\,x}=-\partial_x\frac{\text{Log}\left[-\frac{b\,c-a\,d}{d\,(a+b\,x)}\right]}{g}$
- **Basis:** $\partial_x\left(A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)=\frac{B\,n\,(b\,c-a\,d)}{(a+b\,x)\,(c+d\,x)}$
- **Rule:** If $b\,c-a\,d\neq 0\,\wedge\,b\,f-a\,g=0$, then

$$\int \frac{A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]}{f+g\,x} dx \longrightarrow -\frac{\text{Log}\left[-\frac{b\,c-a\,d}{d\,(a+b\,x)}\right]\left(A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)}{g}+\frac{B\,n\,(b\,c-a\,d)}{g}\int \frac{\text{Log}\left[-\frac{b\,c-a\,d}{d\,(a+b\,x)}\right]}{(a+b\,x)\,(c+d\,x)} dx$$

- **Program code:**

```
Int[(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])/(f_.+g_.*x_),x_Symbol] :=
  -Log[-(b*c-a*d)/(d*(a+b*x))]*(A+B*Log[e*((a+b*x)/(c+d*x))^n])/g +
  B*n*(b*c-a*d)/g*Int[Log[-(b*c-a*d)/(d*(a+b*x))]/((a+b*x)*(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f,g,A,B,n},x] && NeQ[b*c-a*d,0] && EqQ[b*f-a*g,0]
```

```
Int[(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])/(f_.+g_.*x_),x_Symbol] :=
  -Log[-(b*c-a*d)/(d*(a+b*x))]*(A+B*Log[e*(a+b*x)^n/(c+d*x)^n])/g +
  B*n*(b*c-a*d)/g*Int[Log[-(b*c-a*d)/(d*(a+b*x))]/((a+b*x)*(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f,g,A,B,n},x] && EqQ[n+mn,0] && NeQ[b*c-a*d,0] && EqQ[b*f-a*g,0]
```

2: $\int \frac{A + B \operatorname{Log}\left[e \left(\frac{a+b x}{c+d x}\right)^n\right]}{f + g x} \, dx$ when $b c - a d \neq 0 \land d f - c g == 0$

Derivation: Integration by parts

- **Basis: If** $d f - c g == 0$, **then** $\frac{1}{f+g x} == -\partial_x \frac{\operatorname{Log}\left[\frac{b c-a d}{b (c+d x)}\right]}{g}$

- **Basis:** $\partial_x \left(A + B \operatorname{Log}\left[e \left(\frac{a+b x}{c+d x}\right)^n\right]\right) == \frac{B n (b c-a d)}{(a+b x) (c+d x)}$

Rule: If $b c - a d \neq 0 \land d f - c g == 0$, **then**

$$\int \frac{A + B \operatorname{Log}\left[e \left(\frac{a+b x}{c+d x}\right)^n\right]}{f + g x} \, dx \longrightarrow -\frac{\operatorname{Log}\left[\frac{b c-a d}{b (c+d x)}\right] \left(A + B \operatorname{Log}\left[e \left(\frac{a+b x}{c+d x}\right)^n\right]\right)}{g} + \frac{B n (b c - a d)}{g} \int \frac{\operatorname{Log}\left[\frac{b c-a d}{b (c+d x)}\right]}{(a + b x) (c + d x)} \, dx$$

Program code:

```
Int[(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])/(f_.+g_.*x_),x_Symbol] :=
  -Log[(b*c-a*d)/(b*(c+d*x))]*(A+B*Log[e*((a+b*x)/(c+d*x))^n])/g +
  B*n*(b*c-a*d)/g*Int[Log[(b*c-a*d)/(b*(c+d*x))]/((a+b*x)*(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f,g,A,B,n},x] && NeQ[b*c-a*d,0] && EqQ[d*f-c*g,0]
```

```
Int[(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])/(f_.+g_.*x_),x_Symbol] :=
  -Log[(b*c-a*d)/(b*(c+d*x))]*(A+B*Log[e*(a+b*x)^n/(c+d*x)^n])/g +
  B*n*(b*c-a*d)/g*Int[Log[(b*c-a*d)/(b*(c+d*x))]/((a+b*x)*(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f,g,A,B,n},x] && EqQ[n+mn,0] && NeQ[b*c-a*d,0] && EqQ[d*f-c*g,0]
```

3: $\int \frac{A+B\operatorname{Log}\left[e\left(\frac{a+b x}{c+d x}\right)^n\right]}{f+g x}\,dx$ when $b c-a d\neq 0$

Derivation: Integration by parts

Basis: $\partial_x\left(A+B\operatorname{Log}\left[e\left(\frac{a+b x}{c+d x}\right)^n\right]\right) == \frac{b B n}{a+b x}-\frac{B d n}{c+d x}$

Rule: If $b c-a d\neq 0$**, then**

$$\int \frac{A+B\operatorname{Log}\left[e\left(\frac{a+b x}{c+d x}\right)^n\right]}{f+g x}\,dx \longrightarrow \frac{\operatorname{Log}[f+g x]\left(A+B\operatorname{Log}\left[e\left(\frac{a+b x}{c+d x}\right)^n\right]\right)}{g}-\frac{b B n}{g}\int\frac{\operatorname{Log}[f+g x]}{a+b x}\,dx+\frac{B d n}{g}\int\frac{\operatorname{Log}[f+g x]}{c+d x}\,dx$$

Program code:

```
Int[(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])/(f_.+g_.*x_),x_Symbol] :=
  Log[f+g*x]*(A+B*Log[e*((a+b*x)/(c+d*x))^n])/g -
  b*B*n/g*Int[Log[f+g*x]/(a+b*x),x] +
  B*d*n/g*Int[Log[f+g*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,A,B,n},x] && NeQ[b*c-a*d,0]
```

```
Int[(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])/(f_.+g_.*x_),x_Symbol] :=
  Log[f+g*x]*(A+B*Log[e*(a+b*x)^n/(c+d*x)^n])/g -
  b*B*n/g*Int[Log[f+g*x]/(a+b*x),x] +
  B*d*n/g*Int[Log[f+g*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,A,B,n},x] && EqQ[n+mn,0] && NeQ[b*c-a*d,0]
```

2: $\int (f+g\,x)^m \left(A+B\,\mathrm{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right) dx$ when $b\,c-a\,d\neq 0\,\wedge\,m\neq -1\,\wedge\,m\neq -2$

Derivation: Integration by parts

Basis: $\partial_x\left(A+B\,\mathrm{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right) == \frac{B\,n\,(b\,c-a\,d)}{(a+b\,x)\,(c+d\,x)}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,m\neq -1\,\wedge\,m\neq -2$**, then**

$$\int (f+g\,x)^m \left(A+B\,\mathrm{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right) dx \longrightarrow \frac{(f+g\,x)^{m+1}\left(A+B\,\mathrm{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)}{g\,(m+1)}-\frac{B\,n\,(b\,c-a\,d)}{g\,(m+1)}\int\frac{(f+g\,x)^{m+1}}{(a+b\,x)\,(c+d\,x)}\,dx$$

Program code:

```
Int[(f_.+g_.*x_)^m_.*(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.]),x_Symbol] :=
  (f+g*x)^(m+1)*(A+B*Log[e*((a+b*x)/(c+d*x))^n])/(g*(m+1)) -
  B*n*(b*c-a*d)/(g*(m+1))*Int[(f+g*x)^(m+1)/((a+b*x)*(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f,g,A,B,m,n},x] && NeQ[b*c-a*d,0] && NeQ[m,-1] && NeQ[m,-2]
```

```
Int[(f_.+g_.*x_)^m_.*(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_]),x_Symbol] :=
  (f+g*x)^(m+1)*(A+B*Log[e*(a+b*x)^n/(c+d*x)^n])/(g*(m+1)) -
  B*n*(b*c-a*d)/(g*(m+1))*Int[(f+g*x)^(m+1)/((a+b*x)*(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f,g,A,B,m,n},x] && EqQ[n+mn,0] && NeQ[b*c-a*d,0] && NeQ[m,-1] && Not[EqQ[m,-2] && IntegerQ[n]]
```

2. $\int (f+g\,x)^m \left(A+B\,\mathrm{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx$ when $b\,c-a\,d\neq 0\,\wedge\,(m\mid p)\in\mathbb{Z}$

1: $\int (f+g\,x)^m \left(A+B\,\mathrm{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx$ when $b\,c-a\,d\neq 0\,\wedge\,(m\mid p)\in\mathbb{Z}\,\wedge\,b\,f-a\,g==0\,\wedge\,(p>0\vee m<-1)$

- **Derivation: Integration by substitution**

- **Basis:** $F\left[x,\ \frac{a+b\,x}{c+d\,x}\right] == (b\,c-a\,d)\ \mathrm{Subst}\left[\frac{F\left[-\frac{a-c\,x}{b-d\,x},x\right]}{(b-d\,x)^2},\ x,\ \frac{a+b\,x}{c+d\,x}\right]\partial_x\frac{a+b\,x}{c+d\,x}$

- **Rule: If** $b\,c-a\,d\neq 0\,\wedge\,(m\mid p)\in\mathbb{Z}\,\wedge\,b\,f-a\,g==0\,\wedge\,(p>0\vee m<-1)$, **then**

$$\int (f+g\,x)^m \left(A+B\,\mathrm{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx \longrightarrow (b\,c-a\,d)^{m+1}\left(\frac{g}{b}\right)^m \mathrm{Subst}\left[\int\frac{x^m\,(A+B\,\mathrm{Log}[e\,x^n])^p}{(b-d\,x)^{m+2}}dx,\ x,\ \frac{a+b\,x}{c+d\,x}\right]$$

Program code:

```
Int[(f_.+g_.*x_)^m_.*(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])^p_.,x_Symbol] :=
  (b*c-a*d)^(m+1)*(g/b)^m*Subst[Int[x^m*(A+B*Log[e*x^n])^p/(b-d*x)^(m+2),x],x,(a+b*x)/(c+d*x)] /;
FreeQ[{a,b,c,d,e,f,g,A,B,n},x] && NeQ[b*c-a*d,0] && IntegersQ[m,p] && EqQ[b*f-a*g,0] && (GtQ[p,0] || LtQ[m,-1])
```

```
Int[(f_.+g_.*x_)^m_.*(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])^p_.,x_Symbol] :=
  (b*c-a*d)^(m+1)*(g/b)^m*Subst[Int[x^m*(A+B*Log[e*x^n])^p/(b-d*x)^(m+2),x],x,(a+b*x)/(c+d*x)] /;
FreeQ[{a,b,c,d,e,f,g,A,B,n},x] && EqQ[n+mn,0] && IGtQ[n,0] && NeQ[b*c-a*d,0] && IntegersQ[m,p] && EqQ[b*f-a*g,0] && (GtQ[p,0] ||
```

2: $\int (f+g\,x)^m \left(A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx$ when $b\,c-a\,d\neq 0\ \wedge\ (m\mid p)\in\mathbb{Z}\ \wedge\ d\,f-c\,g==0\ \wedge\ (p>0\vee m<-1)$

Derivation: Integration by substitution

- **Basis:** $F\left[x,\frac{a+b\,x}{c+d\,x}\right]==(b\,c-a\,d)\,\text{Subst}\left[\frac{F\left[-\frac{a-c\,x}{b-d\,x},x\right]}{(b-d\,x)^2},\,x,\,\frac{a+b\,x}{c+d\,x}\right]\partial_x\frac{a+b\,x}{c+d\,x}$

Rule: If $b\,c-a\,d\neq 0\ \wedge\ (m\mid p)\in\mathbb{Z}\ \wedge\ d\,f-c\,g==0\ \wedge\ (p>0\vee m<-1)$**, then**

$$\int (f+g\,x)^m \left(A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx \longrightarrow (b\,c-a\,d)^{m+1}\left(\frac{g}{d}\right)^m \text{Subst}\left[\int\frac{(A+B\,\text{Log}[e\,x^n])^p}{(b-d\,x)^{m+2}}dx,\,x,\,\frac{a+b\,x}{c+d\,x}\right]$$

Program code:

```
Int[(f_.+g_.*x_)^m_.*(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])^p_.,x_Symbol] :=
  (b*c-a*d)^(m+1)*(g/d)^m*Subst[Int[(A+B*Log[e*x^n])^p/(b-d*x)^(m+2),x],x,(a+b*x)/(c+d*x)] /;
FreeQ[{a,b,c,d,e,f,g,A,B,n},x] && NeQ[b*c-a*d,0] && IntegersQ[m,p] && EqQ[d*f-c*g,0] && (GtQ[p,0] || LtQ[m,-1])
```

```
Int[(f_.+g_.*x_)^m_.*(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])^p_.,x_Symbol] :=
  (b*c-a*d)^(m+1)*(g/d)^m*Subst[Int[(A+B*Log[e*x^n])^p/(b-d*x)^(m+2),x],x,(a+b*x)/(c+d*x)] /;
FreeQ[{a,b,c,d,e,f,g,A,B,n},x] && EqQ[n+mn,0] && IGtQ[n,0] && NeQ[b*c-a*d,0] && IntegersQ[m,p] && EqQ[d*f-c*g,0] && (GtQ[p,0] ||
```

3: $\int (f+g\,x)^m \left(A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx$ when $b\,c-a\,d\neq 0\ \wedge\ m\in\mathbb{Z}\ \wedge\ p\in\mathbb{Z}^+$

Derivation: Integration by substitution

- **Basis:** $F\left[x,\frac{a+b\,x}{c+d\,x}\right]==(b\,c-a\,d)\,\text{Subst}\left[\frac{F\left[-\frac{a-c\,x}{b-d\,x},x\right]}{(b-d\,x)^2},\,x,\,\frac{a+b\,x}{c+d\,x}\right]\partial_x\frac{a+b\,x}{c+d\,x}$

Rule: If $b\,c-a\,d\neq 0\ \wedge\ m\in\mathbb{Z}\ \wedge\ p\in\mathbb{Z}^+$**, then**

$$\int (f+g\,x)^m \left(A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx \longrightarrow (b\,c-a\,d)\,\text{Subst}\left[\int\frac{(b\,f-a\,g-(d\,f-c\,g)\,x)^m\,(A+B\,\text{Log}[e\,x^n])^p}{(b-d\,x)^{m+2}}dx,\,x,\,\frac{a+b\,x}{c+d\,x}\right]$$

Program code:

```
Int[(f_.+g_.*x_)^m_.*(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])^p_.,x_Symbol] :=
  (b*c-a*d)*Subst[Int[(b*f-a*g-(d*f-c*g)*x)^m*(A+B*Log[e*x^n])^p/(b-d*x)^(m+2),x],x,(a+b*x)/(c+d*x)] /;
FreeQ[{a,b,c,d,e,f,g,A,B,n},x] && NeQ[b*c-a*d,0] && IntegerQ[m] && IGtQ[p,0]
```

```
Int[(f_.+g_.*x_)^m_.*(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])^p_.,x_Symbol] :=
  (b*c-a*d)*Subst[Int[(b*f-a*g-(d*f-c*g)*x)^m*(A+B*Log[e*x^n])^p/(b-d*x)^(m+2),x],x,(a+b*x)/(c+d*x)] /;
FreeQ[{a,b,c,d,e,f,g,A,B,n},x] && EqQ[n+mn,0] && IGtQ[n,0] && NeQ[b*c-a*d,0] && IntegerQ[m] && IGtQ[p,0]
```

U: $\int (f+g\,x)^m \left(A+B\,\mathrm{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx$

- **Rule:**

$$\int (f+g\,x)^m \left(A+B\,\mathrm{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx \longrightarrow \int (f+g\,x)^m \left(A+B\,\mathrm{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx$$

- **Program code:**

```
Int[(f_.+g_.*x_)^m_.*(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])^p_.,x_Symbol] :=
  Unintegrable[(f+g*x)^m*(A+B*Log[e*((a+b*x)/(c+d*x))^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,A,B,m,n,p},x]
```

```
Int[(f_.+g_.*x_)^m_.*(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])^p_.,x_Symbol] :=
  Unintegrable[(f+g*x)^m*(A+B*Log[e*(a+b*x)^n/(c+d*x)^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,A,B,m,n,p},x] && EqQ[n+mn,0] && IntegerQ[n]
```

N: $\int w^m \left(A+B\,\mathrm{Log}\left[e\left(\frac{u}{v}\right)^n\right]\right)^p dx$ when $u == a+b\,x \wedge v == c+d\,x \wedge w == f+g\,x$

- **Derivation: Algebraic normalization**
- **Rule: If** $u == a+b\,x \wedge v == c+d\,x \wedge w == f+g\,x$, **then**

$$\int w^m \left(A+B\,\mathrm{Log}\left[e\left(\frac{u}{v}\right)^n\right]\right)^p dx \longrightarrow \int (f+g\,x)^m \left(A+B\,\mathrm{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx$$

- **Program code:**

```
Int[w_^m_.*(A_.+B_.*Log[e_.*(u_/v_)^n_.])^p_.,x_Symbol] :=
  Int[ExpandToSum[w,x]^m*(A+B*Log[e*(ExpandToSum[u,x]/ExpandToSum[v,x])^n])^p,x] /;
FreeQ[{e,A,B,m,n,p},x] && LinearQ[{u,v,w},x] && Not[LinearMatchQ[{u,v,w},x]]
```

```
Int[w_^m_.*(A_.+B_.*Log[e_.*u_^n_.*v_^mn_])^p_.,x_Symbol] :=
  Int[ExpandToSum[w,x]^m*(A+B*Log[e*ExpandToSum[u,x]^n/ExpandToSum[v,x]^n])^p,x] /;
FreeQ[{e,A,B,m,n,p},x] && EqQ[n+mn,0] && IGtQ[n,0] && LinearQ[{u,v,w},x] && Not[LinearMatchQ[{u,v,w},x]]
```